squashed?

.gitattributes

@@ -49,3 +49,4 @@ saved_model/**/* filter=lfs diff=lfs merge=lfs -text
 *.jpg filter=lfs diff=lfs merge=lfs -text
 *.jpeg filter=lfs diff=lfs merge=lfs -text
 *.webp filter=lfs diff=lfs merge=lfs -text
+formal/setmm/set.mm filter=lfs diff=lfs merge=lfs -text

fetch_books_and_formal.py

@@ -17,6 +17,18 @@ PROOFWIKI_URL = (
     "https://zenodo.org/record/4902289/files/naturalproofs_proofwiki.json?download=1"
 )
 
+def check_encoding(path): 
+    for f in os.listdir(path): 
+        f_path = os.path.join(path, f)
+        if os.path.isfile(f_path): 
+            with open(f_path, encoding="utf-8") as fle: 
+                try: 
+                    fle.read()
+                except UnicodeDecodeError: 
+                    print(f"{f_path} is not unicode")
+        elif os.path.isdir(f_path): 
+            check_encoding(f_path)
+
 
 def _get_dir_from_repo(author, repo, sha, repo_dir, save_path, creds):
     """
@@ -313,6 +325,8 @@ def hol(testing=False):
         if os.path.isfile(f_path):
             os.remove(f_path)
 
+    os.system("rm -r formal/hol/Proofrecording")
+
     _delete_files_except_pattern(save_dir, r".*\.ml|.*\.doc")
 
 
@@ -560,7 +574,7 @@ def main():
     coq(creds)
     lean(creds)
     hol()
-    cam()
+    #cam()
 
 
 if __name__ == "__main__":

fetch_mathoverflow.py

@@ -24,6 +24,14 @@ a structured set of questions with answers.
 3. Run `questions()` and run it to get a dictionary of mathoverflow questions.
    Each question has an `Answers` field that contains a list of answers for the given q.
 """
+def batch_loader(seq, size):
+    """
+    Iterator that takes in a list `seq` and returns
+    chunks of size `size` 
+    """
+    return [seq[pos:pos + size] for pos in range(0, len(seq), size)]
+
+DOC_SEP = "<|endoftext|>"
 
 # source: https://meta.stackexchange.com/questions/2677/database-schema-documentation-for-the-public-data-dump-and-sede
 class PostType(Enum):
@@ -180,16 +188,26 @@ def get_and_format(url, save_dir):
     print("parsing xml...")
     qs = questions()
 
+    print("converting xml to text...")
     qs_texts = [text_of_post(qs[key]) for key in tqdm(qs.keys())]
 
-    for post, score, eyed, answered in tqdm(qs_texts): 
-        if score >= 5 and answered: 
-            with open(os.path.join(save_dir, str(eyed) + ".txt"), "w") as f: 
-                f.write(post)
+    batches = batch_loader(qs_texts, 5000)
+    
+    for i, batch in tqdm(enumerate(batches)): 
+        shard_path = os.path.join(save_dir, f"shard_{i}.txt")
+
+        to_cat = [post for post, score, _, answered in batch
+                if score >=5 and answered]
+        shard = f"{DOC_SEP}\n".join(to_cat)
+
+        with open(shard_path, "w") as f: 
+            f.write(shard)
 
     os.system(f"rm -r {DATA_DIR}")
     os.remove(archive_path)
 
 if __name__ == '__main__':
     get_and_format("https://archive.org/download/stackexchange/mathoverflow.net.7z", 
-            "stack_exchange/math_overflow")
+            "stack-exchange/math_overflow")
+    get_and_format("https://archive.org/download/stackexchange/math.stackexchange.com.7z", 
+            "stack-exchange/math_stack_exchange")

formal/afp/ADS_Functor/ADS_Construction.thy

@@ -0,0 +1,1281 @@
+(* Author: Andreas Lochbihler, Digital Asset
+   Author: Ognjen Maric, Digital Asset *)
+
+theory ADS_Construction imports
+  Merkle_Interface
+  "HOL-Library.Simps_Case_Conv"
+begin
+
+(************************************************************)
+section \<open> Building blocks for authenticated data structures on datatypes \<close>
+(************************************************************)
+
+(************************************************************)
+subsection \<open> Building Block: Identity Functor \<close>
+(************************************************************)
+
+text \<open>If nothing is blindable in a type, then the type itself is the hash and the ADS of itself.\<close>
+
+abbreviation (input) hash_discrete :: "('a, 'a) hash" where "hash_discrete \<equiv> id"
+
+abbreviation (input) blinding_of_discrete :: "'a blinding_of" where
+  "blinding_of_discrete \<equiv> (=)"
+
+definition merge_discrete :: "'a merge" where
+  "merge_discrete x y = (if x = y then Some y else None)"
+
+lemma blinding_of_discrete_hash:
+  "blinding_of_discrete \<le> vimage2p hash_discrete hash_discrete (=)"
+  by(auto simp add: vimage2p_def)
+
+lemma blinding_of_on_discrete [locale_witness]:
+  "blinding_of_on UNIV hash_discrete blinding_of_discrete"
+  by(unfold_locales)(simp_all add: OO_eq eq_onp_def blinding_of_discrete_hash)
+
+lemma merge_on_discrete [locale_witness]:
+  "merge_on UNIV hash_discrete blinding_of_discrete merge_discrete"
+  by unfold_locales(auto simp add: merge_discrete_def)
+
+lemma merkle_discrete [locale_witness]:
+  "merkle_interface hash_discrete blinding_of_discrete merge_discrete"
+  ..
+
+parametric_constant merge_discrete_parametric [transfer_rule]: merge_discrete_def
+
+(************************************************************)
+subsubsection \<open>Example: instantiation for @{typ unit}\<close>
+(************************************************************)
+
+abbreviation (input) hash_unit :: "(unit, unit) hash" where "hash_unit \<equiv> hash_discrete"
+
+abbreviation blinding_of_unit :: "unit blinding_of" where
+  "blinding_of_unit \<equiv> blinding_of_discrete"
+
+abbreviation merge_unit :: "unit merge" where "merge_unit \<equiv> merge_discrete"
+
+lemma blinding_of_unit_hash:
+  "blinding_of_unit \<le> vimage2p hash_unit hash_unit (=)"
+  by(fact blinding_of_discrete_hash)
+
+lemma blinding_of_on_unit:
+  "blinding_of_on UNIV hash_unit blinding_of_unit"
+  by(fact blinding_of_on_discrete)
+
+lemma merge_on_unit:
+  "merge_on UNIV hash_unit blinding_of_unit merge_unit"
+  by(fact merge_on_discrete)
+
+lemma merkle_interface_unit:
+  "merkle_interface hash_unit blinding_of_unit merge_unit"
+  by(intro merkle_interfaceI merge_on_unit)
+
+(************************************************************)
+subsection \<open> Building Block: Blindable Position \<close>
+(************************************************************)
+
+type_synonym 'a blindable = 'a
+
+text \<open> The following type represents the hashes of a datatype. We model hashes as being injective,
+  but not surjective; some hashes do not correspond to any values of the original datatypes. We
+  model such values as "garbage" coming from a countable set (here, naturals). \<close>
+
+type_synonym garbage = nat
+
+datatype 'a\<^sub>h blindable\<^sub>h = Content 'a\<^sub>h | Garbage garbage
+
+datatype ('a\<^sub>m, 'a\<^sub>h) blindable\<^sub>m = Unblinded 'a\<^sub>m | Blinded "'a\<^sub>h blindable\<^sub>h"
+
+(************************************************************)
+subsubsection \<open> Hashes \<close>
+(************************************************************)
+
+primrec hash_blindable' :: "(('a\<^sub>h, 'a\<^sub>h) blindable\<^sub>m, 'a\<^sub>h blindable\<^sub>h) hash" where
+  "hash_blindable' (Unblinded x) = Content x"
+| "hash_blindable' (Blinded x) = x"
+
+definition hash_blindable :: "('a\<^sub>m, 'a\<^sub>h) hash \<Rightarrow> (('a\<^sub>m, 'a\<^sub>h) blindable\<^sub>m, 'a\<^sub>h blindable\<^sub>h) hash" where
+  "hash_blindable h = hash_blindable' \<circ> map_blindable\<^sub>m h id"
+
+lemma hash_blindable_simps [simp]:
+  "hash_blindable h (Unblinded x) = Content (h x)"
+  "hash_blindable h (Blinded y) = y"
+  by(simp_all add: hash_blindable_def blindable\<^sub>h.map_id)
+
+lemma hash_map_blindable_simp:
+  "hash_blindable f (map_blindable\<^sub>m f' id x) = hash_blindable (f o f') x"
+  by(cases x) (simp_all add: hash_blindable_def blindable\<^sub>h.map_comp)
+
+parametric_constant hash_blindable'_parametric [transfer_rule]: hash_blindable'_def
+
+parametric_constant hash_blindable_parametric [transfer_rule]: hash_blindable_def
+
+(************************************************************)
+subsubsection \<open> Blinding \<close>
+(************************************************************)
+
+context
+  fixes h :: "('a\<^sub>m, 'a\<^sub>h) hash"
+    and bo :: "'a\<^sub>m blinding_of"
+begin
+
+inductive blinding_of_blindable :: "('a\<^sub>m, 'a\<^sub>h) blindable\<^sub>m blinding_of" where
+  "blinding_of_blindable (Unblinded x) (Unblinded y)" if "bo x y"
+| "blinding_of_blindable (Blinded x) t" if "hash_blindable h t = x"
+
+inductive_simps blinding_of_blindable_simps [simp]:
+  "blinding_of_blindable (Unblinded x) y"
+  "blinding_of_blindable (Blinded x) y"
+  "blinding_of_blindable z (Unblinded x)"
+  "blinding_of_blindable z (Blinded x)"
+
+inductive_simps blinding_of_blindable_simps2:
+   "blinding_of_blindable (Unblinded x) (Unblinded y)"
+   "blinding_of_blindable (Unblinded x) (Blinded y')"
+   "blinding_of_blindable (Blinded x') (Unblinded y)"
+   "blinding_of_blindable (Blinded x') (Blinded y')"
+
+end
+
+lemma blinding_of_blindable_mono:
+  assumes "bo \<le> bo'"
+  shows "blinding_of_blindable h bo \<le> blinding_of_blindable h bo'"
+  apply(rule predicate2I)
+  apply(erule blinding_of_blindable.cases; hypsubst)
+  subgoal by(rule blinding_of_blindable.intros)(rule assms[THEN predicate2D])
+  subgoal by(rule blinding_of_blindable.intros) simp
+  done
+
+lemma blinding_of_blindable_hash:
+  assumes "bo \<le> vimage2p h h (=)"
+  shows "blinding_of_blindable h bo \<le> vimage2p (hash_blindable h) (hash_blindable h) (=)"
+  apply(rule predicate2I vimage2pI)+
+  apply(erule blinding_of_blindable.cases; hypsubst)
+  subgoal using assms[THEN predicate2D] by(simp add: vimage2p_def)
+  subgoal by simp
+  done
+
+lemma blinding_of_on_blindable [locale_witness]:
+  assumes "blinding_of_on A h bo"
+  shows "blinding_of_on {x. set1_blindable\<^sub>m x \<subseteq> A} (hash_blindable h) (blinding_of_blindable h bo)"
+  (is "blinding_of_on ?A ?h ?bo")
+proof -
+  interpret blinding_of_on A h bo by fact
+  show ?thesis
+  proof
+    show "?bo \<le> vimage2p ?h ?h (=)"
+      by(rule blinding_of_blindable_hash)(rule hash)
+    show "?bo x x" if "x \<in> ?A" for x using that by(cases x)(auto simp add: refl)
+    show "?bo x z" if "?bo x y" "?bo y z" "x \<in> ?A" for x y z using that
+      by(auto elim!: blinding_of_blindable.cases dest: trans blinding_hash_eq)
+    show "x = y" if "?bo x y" "?bo y x" "x \<in> ?A" for x y using that
+      by(auto elim!: blinding_of_blindable.cases dest: antisym)
+  qed
+qed
+
+lemmas blinding_of_blindable [locale_witness] = blinding_of_on_blindable[of UNIV, simplified]
+
+case_of_simps blinding_of_blindable_alt_def: blinding_of_blindable_simps2
+parametric_constant blinding_of_blindable_parametric [transfer_rule]: blinding_of_blindable_alt_def
+
+(************************************************************)
+subsubsection \<open> Merging \<close>
+(************************************************************)
+
+context
+  fixes h :: "('a\<^sub>m, 'a\<^sub>h) hash"
+  fixes m :: "'a\<^sub>m merge"
+begin
+
+fun merge_blindable :: "('a\<^sub>m, 'a\<^sub>h) blindable\<^sub>m merge" where
+  "merge_blindable (Unblinded x) (Unblinded y) = map_option Unblinded (m x y)"
+| "merge_blindable (Blinded x) (Unblinded y) = (if x = Content (h y) then Some (Unblinded y) else None)"
+| "merge_blindable (Unblinded y) (Blinded x) = (if x = Content (h y) then Some (Unblinded y) else None)"
+| "merge_blindable (Blinded t) (Blinded u) = (if t = u then Some (Blinded u) else None)"
+
+lemma merge_on_blindable [locale_witness]:
+  assumes "merge_on A h bo m"
+  shows "merge_on {x. set1_blindable\<^sub>m x \<subseteq> A} (hash_blindable h) (blinding_of_blindable h bo) merge_blindable"
+  (is "merge_on ?A ?h ?bo ?m")
+proof -
+  interpret merge_on A h bo m by fact
+  show ?thesis
+  proof
+    show "\<exists>ab. ?m a b = Some ab \<and> ?bo a ab \<and> ?bo b ab \<and> (\<forall>u. ?bo a u \<longrightarrow> ?bo b u \<longrightarrow> ?bo ab u)" if "?h a = ?h b" "a \<in> ?A" for a b
+      using that by(cases "(a, b)" rule: merge_blindable.cases)(auto simp add: refl dest!: join)
+    show "?m a b = None" if "?h a \<noteq> ?h b" "a \<in> ?A" for a b 
+      using that by(cases "(a, b)" rule: merge_blindable.cases)(auto simp add: dest!: undefined)
+  qed
+qed
+
+lemmas merge_blindable [locale_witness] = 
+  merge_on_blindable[of UNIV, simplified]
+
+end
+
+lemma merge_blindable_alt_def:
+  "merge_blindable h m x y = (case (x, y) of
+    (Unblinded x, Unblinded y) \<Rightarrow> map_option Unblinded (m x y)
+  | (Blinded x, Unblinded y) \<Rightarrow> (if Content (h y) = x then Some (Unblinded y) else None)
+  | (Unblinded y, Blinded x) \<Rightarrow> (if Content (h y) = x then Some (Unblinded y) else None)
+  | (Blinded t, Blinded u) \<Rightarrow> (if t = u then Some (Blinded u) else None))"
+  by(simp split: blindable\<^sub>m.split blindable\<^sub>h.split)
+
+parametric_constant merge_blindable_parametric [transfer_rule]: merge_blindable_alt_def
+
+lemma merge_blindable_cong [fundef_cong]:
+  assumes "\<And>a b. \<lbrakk> a \<in> set1_blindable\<^sub>m x; b \<in> set1_blindable\<^sub>m y \<rbrakk> \<Longrightarrow> m a b = m' a b"
+  shows "merge_blindable h m x y = merge_blindable h m' x y"
+  by(auto simp add: merge_blindable_alt_def split: blindable\<^sub>m.split intro: assms intro!: arg_cong[where f="map_option _"])
+
+(************************************************************)
+subsubsection \<open> Merkle interface \<close>
+(************************************************************)
+
+lemma merkle_blindable [locale_witness]:
+  assumes "merkle_interface h bo m"
+  shows "merkle_interface (hash_blindable h) (blinding_of_blindable h bo) (merge_blindable h m)"
+proof -
+  interpret merge_on UNIV h bo m using assms by(simp add: merkle_interface_aux)
+  show ?thesis unfolding merkle_interface_aux ..
+qed
+
+
+(************************************************************)
+subsubsection \<open> Non-recursive blindable positions \<close>
+(************************************************************)
+
+text \<open> For a non-recursive data type @{typ 'a}, the type of hashes in @{type blindable\<^sub>m} is fixed
+to be simply @{typ "'a blindable\<^sub>h"}. We obtain this by instantiating the type variable with the
+identity building block. \<close>
+
+type_synonym 'a nr_blindable = "('a, 'a) blindable\<^sub>m"
+
+abbreviation hash_nr_blindable :: "('a nr_blindable, 'a blindable\<^sub>h) hash" where
+  "hash_nr_blindable \<equiv> hash_blindable hash_discrete"
+
+abbreviation blinding_of_nr_blindable :: "'a nr_blindable blinding_of" where
+  "blinding_of_nr_blindable \<equiv> blinding_of_blindable hash_discrete blinding_of_discrete"
+
+abbreviation merge_nr_blindable :: "'a nr_blindable merge" where
+  "merge_nr_blindable \<equiv> merge_blindable hash_discrete merge_discrete"
+
+lemma merge_on_nr_blindable:
+  "merge_on UNIV hash_nr_blindable blinding_of_nr_blindable merge_nr_blindable"
+  ..
+
+lemma merkle_nr_blindable:
+  "merkle_interface hash_nr_blindable blinding_of_nr_blindable merge_nr_blindable"
+  ..
+
+(************************************************************)
+subsection \<open> Building block: Sums \<close>
+(************************************************************)
+
+text \<open> We prove that we can lift the ADS construction through sums.\<close>
+
+type_synonym ('a\<^sub>h, 'b\<^sub>h) sum\<^sub>h = "'a\<^sub>h + 'b\<^sub>h"
+type_notation sum\<^sub>h (infixr "+\<^sub>h" 10)
+
+type_synonym ('a\<^sub>m, 'b\<^sub>m) sum\<^sub>m = "'a\<^sub>m + 'b\<^sub>m"
+  \<comment> \<open>If a functor does not introduce blindable positions, then we don't need the type variable copies.\<close>
+type_notation sum\<^sub>m (infixr "+\<^sub>m" 10)
+
+(************************************************************)
+subsubsection \<open> Hashes \<close>
+(************************************************************)
+
+abbreviation (input) hash_sum' :: "('a\<^sub>h +\<^sub>h 'b\<^sub>h, 'a\<^sub>h +\<^sub>h 'b\<^sub>h) hash" where
+  "hash_sum' \<equiv> id"
+
+abbreviation (input) hash_sum :: "('a\<^sub>m, 'a\<^sub>h) hash \<Rightarrow> ('b\<^sub>m, 'b\<^sub>h) hash \<Rightarrow> ('a\<^sub>m +\<^sub>m 'b\<^sub>m, 'a\<^sub>h +\<^sub>h 'b\<^sub>h) hash"
+  where "hash_sum \<equiv> map_sum"
+
+(************************************************************)
+subsubsection \<open> Blinding \<close>
+(************************************************************)
+
+abbreviation (input) blinding_of_sum :: "'a\<^sub>m blinding_of \<Rightarrow> 'b\<^sub>m blinding_of \<Rightarrow> ('a\<^sub>m +\<^sub>m 'b\<^sub>m) blinding_of" where
+  "blinding_of_sum \<equiv> rel_sum"
+
+lemmas blinding_of_sum_mono = sum.rel_mono
+
+lemma blinding_of_sum_hash:
+  assumes "boa \<le> vimage2p rha rha (=)" "bob \<le> vimage2p rhb rhb (=)"
+  shows "blinding_of_sum boa bob \<le> vimage2p (hash_sum rha rhb) (hash_sum rha rhb) (=)"
+  using assms by(auto simp add: vimage2p_def elim!: rel_sum.cases)
+
+lemma blinding_of_on_sum [locale_witness]:
+  assumes "blinding_of_on A rha boa" "blinding_of_on B rhb bob"
+  shows "blinding_of_on {x. setl x \<subseteq> A \<and> setr x \<subseteq> B} (hash_sum rha rhb) (blinding_of_sum boa bob)"
+  (is "blinding_of_on ?A ?h ?bo")
+proof -
+  interpret a: blinding_of_on A rha boa by fact
+  interpret b: blinding_of_on B rhb bob by fact
+  show ?thesis
+  proof
+    show "?bo x x" if "x \<in> ?A" for x using that by(intro sum.rel_refl_strong)(auto intro: a.refl b.refl)
+    show "?bo x z" if "?bo x y" "?bo y z" "x \<in> ?A" for x y z
+      using that by(auto elim!: rel_sum.cases dest: a.trans b.trans)
+    show "x = y" if "?bo x y" "?bo y x" "x \<in> ?A" for x y 
+      using that by(auto elim!: rel_sum.cases dest: a.antisym b.antisym)
+  qed(rule blinding_of_sum_hash a.hash b.hash)+
+qed
+
+lemmas blinding_of_sum [locale_witness] = blinding_of_on_sum[of UNIV _ _ UNIV, simplified]
+
+(************************************************************)
+subsubsection \<open> Merging \<close>
+(************************************************************)
+
+context
+  fixes ma :: "'a\<^sub>m merge"
+  fixes mb :: "'b\<^sub>m merge"
+begin
+
+fun merge_sum :: "('a\<^sub>m +\<^sub>m 'b\<^sub>m) merge" where
+  "merge_sum (Inl x) (Inl y) = map_option Inl (ma x y)"
+| "merge_sum (Inr x) (Inr y) = map_option Inr (mb x y)"
+| "merge_sum _ _ = None"
+
+lemma merge_on_sum [locale_witness]:
+  assumes "merge_on A rha boa ma" "merge_on B rhb bob mb"
+  shows "merge_on {x. setl x \<subseteq> A \<and> setr x \<subseteq> B} (hash_sum rha rhb) (blinding_of_sum boa bob) merge_sum"
+  (is "merge_on ?A ?h ?bo ?m")
+proof -
+  interpret a: merge_on A rha boa ma by fact
+  interpret b: merge_on B rhb bob mb by fact
+  show ?thesis
+  proof
+    show "\<exists>ab. ?m a b = Some ab \<and> ?bo a ab \<and> ?bo b ab \<and> (\<forall>u. ?bo a u \<longrightarrow> ?bo b u \<longrightarrow> ?bo ab u)"
+      if "?h a = ?h b" "a \<in> ?A" for a b using that
+      by(cases "(a, b)" rule: merge_sum.cases)(auto dest!: a.join b.join elim!: rel_sum.cases)
+    show "?m a b = None" if "?h a \<noteq> ?h b" "a \<in> ?A" for a b using that
+      by(cases "(a, b)" rule: merge_sum.cases)(auto dest!: a.undefined b.undefined)
+  qed
+qed
+
+lemmas merge_sum [locale_witness] = merge_on_sum[where A=UNIV and B=UNIV, simplified]
+
+lemma merge_sum_alt_def:
+  "merge_sum x y = (case (x, y) of
+    (Inl x, Inl y) \<Rightarrow> map_option Inl (ma x y)
+  | (Inr x, Inr y) \<Rightarrow> map_option Inr (mb x y)
+  | _ \<Rightarrow> None)"
+  by(simp add: split: sum.split)
+
+end
+
+lemma merge_sum_cong[fundef_cong]:
+  "\<lbrakk> x = x'; y = y'; 
+    \<And>xl yl. \<lbrakk> x = Inl xl; y = Inl yl \<rbrakk> \<Longrightarrow> ma xl yl = ma' xl yl;
+    \<And>xr yr. \<lbrakk> x = Inr xr; y = Inr yr \<rbrakk> \<Longrightarrow> mb xr yr = mb' xr yr \<rbrakk> \<Longrightarrow>
+    merge_sum ma mb x y = merge_sum ma' mb' x' y'"
+  by(cases x; simp_all; cases y; auto)
+
+parametric_constant merge_sum_parametric [transfer_rule]: merge_sum_alt_def
+
+subsubsection \<open> Merkle interface \<close>
+
+lemma merkle_sum [locale_witness]:
+  assumes "merkle_interface rha boa ma" "merkle_interface rhb bob mb"
+  shows "merkle_interface (hash_sum rha rhb) (blinding_of_sum boa bob) (merge_sum ma mb)"
+proof -
+  interpret a: merge_on UNIV rha boa ma unfolding merkle_interface_aux[symmetric] by fact
+  interpret b: merge_on UNIV rhb bob mb unfolding merkle_interface_aux[symmetric] by fact
+  show ?thesis unfolding merkle_interface_aux[symmetric] ..
+qed
+
+(************************************************************)
+subsection \<open> Building Block: Products\<close>
+(************************************************************)
+
+text \<open> We prove that we can lift the ADS construction through products.\<close>
+
+type_synonym ('a\<^sub>h, 'b\<^sub>h) prod\<^sub>h = "'a\<^sub>h \<times> 'b\<^sub>h"
+type_notation prod\<^sub>h ("(_ \<times>\<^sub>h/ _)" [21, 20] 20)
+
+type_synonym ('a\<^sub>m, 'b\<^sub>m) prod\<^sub>m = "'a\<^sub>m \<times> 'b\<^sub>m"
+  \<comment> \<open>If a functor does not introduce blindable positions, then we don't need the type variable copies.\<close>
+type_notation prod\<^sub>m ("(_ \<times>\<^sub>m/ _)" [21, 20] 20)
+
+(************************************************************)
+subsubsection \<open> Hashes \<close>
+(************************************************************)
+
+abbreviation (input) hash_prod' :: "('a\<^sub>h \<times>\<^sub>h 'b\<^sub>h, 'a\<^sub>h \<times>\<^sub>h 'b\<^sub>h) hash" where
+  "hash_prod' \<equiv> id"
+
+abbreviation (input) hash_prod :: "('a\<^sub>m, 'a\<^sub>h) hash \<Rightarrow> ('b\<^sub>m, 'b\<^sub>h) hash \<Rightarrow> ('a\<^sub>m \<times>\<^sub>m 'b\<^sub>m, 'a\<^sub>h \<times>\<^sub>h 'b\<^sub>h) hash"
+  where "hash_prod \<equiv> map_prod"
+
+(************************************************************)
+subsubsection \<open> Blinding \<close>
+(************************************************************)
+
+abbreviation (input) blinding_of_prod :: "'a\<^sub>m blinding_of \<Rightarrow> 'b\<^sub>m blinding_of \<Rightarrow> ('a\<^sub>m \<times>\<^sub>m 'b\<^sub>m) blinding_of" where
+  "blinding_of_prod \<equiv> rel_prod"
+
+lemmas blinding_of_prod_mono = prod.rel_mono
+
+lemma blinding_of_prod_hash:
+  assumes "boa \<le> vimage2p rha rha (=)" "bob \<le> vimage2p rhb rhb (=)"
+  shows "blinding_of_prod boa bob \<le> vimage2p (hash_prod rha rhb) (hash_prod rha rhb) (=)"
+  using assms by(auto simp add: vimage2p_def)
+
+lemma blinding_of_on_prod [locale_witness]:
+  assumes "blinding_of_on A rha boa" "blinding_of_on B rhb bob"
+  shows "blinding_of_on {x. fsts x \<subseteq> A \<and> snds x \<subseteq> B} (hash_prod rha rhb) (blinding_of_prod boa bob)"
+  (is "blinding_of_on ?A ?h ?bo")
+proof -
+  interpret a: blinding_of_on A rha boa by fact
+  interpret b: blinding_of_on B rhb bob by fact
+  show ?thesis
+  proof
+    show "?bo x x" if "x \<in> ?A" for x using that by(cases x)(auto intro: a.refl b.refl)
+    show "?bo x z" if "?bo x y" "?bo y z" "x \<in> ?A" for x y z using that
+      by(auto elim!: rel_prod.cases dest: a.trans b.trans)
+    show "x = y" if "?bo x y" "?bo y x" "x \<in> ?A" for x y using that
+      by(auto elim!: rel_prod.cases dest: a.antisym b.antisym)
+  qed(rule blinding_of_prod_hash a.hash b.hash)+
+qed
+
+lemmas blinding_of_prod [locale_witness] = blinding_of_on_prod[where A=UNIV and B=UNIV, simplified]
+
+(************************************************************)
+subsubsection \<open> Merging \<close>
+(************************************************************)
+
+context
+  fixes ma :: "'a\<^sub>m merge"
+  fixes mb :: "'b\<^sub>m merge"
+begin
+
+fun merge_prod :: "('a\<^sub>m \<times>\<^sub>m 'b\<^sub>m) merge" where
+  "merge_prod (x, y) (x', y') = Option.bind (ma x x') (\<lambda>x''. map_option (Pair x'') (mb y y'))"
+
+lemma merge_on_prod [locale_witness]:
+  assumes "merge_on A rha boa ma" "merge_on B rhb bob mb"
+  shows "merge_on {x. fsts x \<subseteq> A \<and> snds x \<subseteq> B} (hash_prod rha rhb) (blinding_of_prod boa bob) merge_prod"
+  (is "merge_on ?A ?h ?bo ?m")
+proof -
+  interpret a: merge_on A rha boa ma by fact
+  interpret b: merge_on B rhb bob mb by fact
+  show ?thesis
+  proof
+    show "\<exists>ab. ?m a b = Some ab \<and> ?bo a ab \<and> ?bo b ab \<and> (\<forall>u. ?bo a u \<longrightarrow> ?bo b u \<longrightarrow> ?bo ab u)"
+      if "?h a = ?h b" "a \<in> ?A" for a b using that
+      by(cases "(a, b)" rule: merge_prod.cases)(auto dest!: a.join b.join)
+    show "?m a b = None" if "?h a \<noteq> ?h b" "a \<in> ?A" for a b using that
+      by(cases "(a, b)" rule: merge_prod.cases)(auto dest!: a.undefined b.undefined)
+  qed
+qed
+
+lemmas merge_prod [locale_witness] = merge_on_prod[where A=UNIV and B=UNIV, simplified]
+
+lemma merge_prod_alt_def:
+  "merge_prod = (\<lambda>(x, y) (x', y'). Option.bind (ma x x') (\<lambda>x''. map_option (Pair x'') (mb y y')))"
+  by(simp add: fun_eq_iff)
+
+end
+
+lemma merge_prod_cong[fundef_cong]:
+  assumes "\<And>a b. \<lbrakk> a \<in> fsts p1; b \<in> fsts p2 \<rbrakk> \<Longrightarrow> ma a b = ma' a b"
+    and "\<And>a b. \<lbrakk> a \<in> snds p1; b \<in> snds p2 \<rbrakk> \<Longrightarrow> mb a b = mb' a b" 
+  shows "merge_prod ma mb p1 p2 = merge_prod ma' mb' p1 p2"
+  using assms by(cases p1; cases p2) auto
+
+parametric_constant merge_prod_parametric [transfer_rule]: merge_prod_alt_def
+
+(************************************************************)
+subsubsection \<open> Merkle Interface \<close>
+(************************************************************)
+
+lemma merkle_product [locale_witness]:
+  assumes "merkle_interface rha boa ma" "merkle_interface rhb bob mb"
+  shows "merkle_interface (hash_prod rha rhb) (blinding_of_prod boa bob) (merge_prod ma mb)"
+proof -
+  interpret a: merge_on UNIV rha boa ma unfolding merkle_interface_aux[symmetric] by fact
+  interpret b: merge_on UNIV rhb bob mb unfolding merkle_interface_aux[symmetric] by fact
+  show ?thesis unfolding merkle_interface_aux[symmetric] ..
+qed
+
+
+(************************************************************)
+subsection \<open>Building Block: Lists\<close>
+(************************************************************)
+
+text \<open>The ADS construction on lists is done the easiest through a separate isomorphic datatype
+  that has only a single constructor. We hide this construction in a locale. \<close>
+
+locale list_R1 begin
+
+type_synonym ('a, 'b) list_F = "unit + 'a \<times> 'b"
+
+abbreviation (input) "set_base_F\<^sub>m \<equiv> \<lambda>x. setr x \<bind> fsts"
+abbreviation (input) "set_rec_F\<^sub>m \<equiv> \<lambda>A. setr A \<bind> snds"
+abbreviation (input) "map_F \<equiv> \<lambda>fb fr. map_sum id (map_prod fb fr)"
+
+datatype 'a list_R1 = list_R1 (unR: "('a, 'a list_R1) list_F")
+
+lemma list_R1_const_into_dest: "list_R1 F = l \<longleftrightarrow> F = unR l"
+  by auto
+
+declare list_R1.split[split]
+
+lemma list_R1_induct[case_names list_R1]:
+  assumes "\<And>F. \<lbrakk> \<And>l'. l' \<in> set_rec_F\<^sub>m F  \<Longrightarrow> P l' \<rbrakk> \<Longrightarrow> P (list_R1 F)"
+  shows "P l"  
+  apply(rule list_R1.induct)
+  apply(auto intro!: assms)
+  done
+
+lemma set_list_R1_eq: 
+  "{x. set_base_F\<^sub>m x \<subseteq> A \<and> set_rec_F\<^sub>m x \<subseteq> B} =
+   {x. setl x \<subseteq> UNIV \<and> setr x \<subseteq> {x. fsts x \<subseteq> A \<and> snds x \<subseteq> B}}"
+  by(auto simp add: bind_UNION)
+
+(************************************************************)
+subsubsection \<open> The Isomorphism \<close>
+(************************************************************)
+
+primrec (transfer) list_R1_to_list :: "'a list_R1 \<Rightarrow> 'a list" where
+  "list_R1_to_list (list_R1 l) = (case map_sum id (map_prod id list_R1_to_list) l of Inl () \<Rightarrow> [] | Inr (x, xs) \<Rightarrow> x # xs)"
+
+lemma list_R1_to_list_simps [simp]:
+  "list_R1_to_list (list_R1 (Inl ())) = []"
+  "list_R1_to_list (list_R1 (Inr (x, xs))) = x # list_R1_to_list xs"
+  by(simp_all split: unit.split)
+
+declare list_R1_to_list.simps [simp del]
+
+primrec (transfer) list_to_list_R1 :: "'a list \<Rightarrow> 'a list_R1" where
+  "list_to_list_R1 [] = list_R1 (Inl ())"
+| "list_to_list_R1 (x#xs) = list_R1 (Inr (x, list_to_list_R1 xs))"
+
+lemma R1_of_list: "list_R1_to_list (list_to_list_R1 x) = x"
+  by(induct x) (auto)
+
+lemma list_of_R1: "list_to_list_R1 (list_R1_to_list x) = x"
+  apply(induct x)
+  subgoal for x
+    by(cases x) (auto)
+  done
+
+lemma list_R1_def: "type_definition list_to_list_R1 list_R1_to_list UNIV"
+  by(unfold_locales)(auto intro: R1_of_list list_of_R1)
+
+setup_lifting list_R1_def
+
+lemma map_list_R1_list_to_list_R1: "map_list_R1 f (list_to_list_R1 xs) = list_to_list_R1 (map f xs)"
+  by(induction xs) auto
+
+lemma list_R1_map_trans [transfer_rule]: includes lifting_syntax shows
+  "(((=) ===> (=)) ===> pcr_list (=) ===> pcr_list (=)) map_list_R1 map"
+  by(auto 4 3 simp add: list.pcr_cr_eq rel_fun_eq cr_list_def map_list_R1_list_to_list_R1)
+
+lemma set_list_R1_list_to_list_R1: "set_list_R1 (list_to_list_R1 xs) = set xs"
+  by(induction xs) auto
+
+lemma list_R1_set_trans [transfer_rule]: includes lifting_syntax shows
+  "(pcr_list (=) ===> (=)) set_list_R1 set"
+  by(auto simp add: list.pcr_cr_eq cr_list_def set_list_R1_list_to_list_R1)
+
+lemma rel_list_R1_list_to_list_R1:
+   "rel_list_R1 R (list_to_list_R1 xs) (list_to_list_R1 ys) \<longleftrightarrow> list_all2 R xs ys"
+  (is "?lhs \<longleftrightarrow> ?rhs")
+proof
+  define xs' and ys' where "xs' = list_to_list_R1 xs" and "ys' = list_to_list_R1 ys"
+  assume "rel_list_R1 R xs' ys'"
+  then have "list_all2 R (list_R1_to_list xs') (list_R1_to_list ys')"
+    by induction(auto elim!: rel_sum.cases)
+  thus ?rhs by(simp add: xs'_def ys'_def R1_of_list)
+next
+  show ?lhs if ?rhs using that by induction auto
+qed
+
+lemma list_R1_rel_trans[transfer_rule]: includes lifting_syntax shows
+  "(((=) ===> (=) ===> (=)) ===> pcr_list (=) ===> pcr_list (=) ===> (=)) rel_list_R1 list_all2"
+  by(auto 4 4 simp add: list.pcr_cr_eq rel_fun_eq cr_list_def rel_list_R1_list_to_list_R1)
+
+(************************************************************)
+subsubsection \<open> Hashes \<close>
+(************************************************************)
+
+type_synonym ('a\<^sub>h, 'b\<^sub>h) list_F\<^sub>h = "unit +\<^sub>h 'a\<^sub>h \<times>\<^sub>h 'b\<^sub>h"
+
+type_synonym ('a\<^sub>m, 'b\<^sub>m) list_F\<^sub>m = "unit +\<^sub>m 'a\<^sub>m \<times>\<^sub>m 'b\<^sub>m"
+
+type_synonym 'a\<^sub>h list_R1\<^sub>h = "'a\<^sub>h list_R1" 
+  \<comment> \<open>In theory, we should define a separate datatype here of the functor @{typ "('a\<^sub>h, _) list_F\<^sub>h"}.
+    We take a shortcut because they're isomorphic.\<close>
+
+type_synonym 'a\<^sub>m list_R1\<^sub>m = "'a\<^sub>m list_R1"
+  \<comment> \<open>In theory, we should define a separate datatype here of the functor @{typ "('a\<^sub>m, _) list_F\<^sub>m"}.
+    We take a shortcut because they're isomorphic.\<close>
+
+definition hash_F :: "('a\<^sub>m, 'a\<^sub>h) hash \<Rightarrow> ('b\<^sub>m, 'b\<^sub>h) hash \<Rightarrow> (('a\<^sub>m, 'b\<^sub>m) list_F\<^sub>m, ('a\<^sub>h, 'b\<^sub>h) list_F\<^sub>h) hash" where 
+  "hash_F h rhL = hash_sum hash_unit (hash_prod h rhL)"
+
+abbreviation (input) hash_R1 :: "('a\<^sub>m, 'a\<^sub>h) hash \<Rightarrow> ('a\<^sub>m list_R1\<^sub>m, 'a\<^sub>h list_R1\<^sub>h) hash" where
+  "hash_R1 \<equiv> map_list_R1"
+
+parametric_constant hash_F_parametric[transfer_rule]: hash_F_def
+
+(************************************************************)
+subsubsection \<open> Blinding \<close>
+(************************************************************)
+
+definition blinding_of_F :: "'a\<^sub>m blinding_of \<Rightarrow> 'b\<^sub>m blinding_of \<Rightarrow> ('a\<^sub>m, 'b\<^sub>m) list_F\<^sub>m blinding_of" where
+  "blinding_of_F bo bL = blinding_of_sum blinding_of_unit (blinding_of_prod bo bL)"
+ 
+abbreviation (input) blinding_of_R1 :: "'a blinding_of \<Rightarrow> 'a list_R1 blinding_of" where
+  "blinding_of_R1 \<equiv> rel_list_R1"
+
+lemma blinding_of_hash_R1:
+  assumes "bo \<le> vimage2p h h (=)"
+  shows "blinding_of_R1 bo \<le> vimage2p (hash_R1 h) (hash_R1 h) (=)"
+  apply(rule predicate2I vimage2pI)+
+  apply(auto simp add: predicate2D_vimage2p[OF assms] elim!: list_R1.rel_induct rel_sum.cases rel_prod.cases)
+  done
+
+lemma blinding_of_on_R1 [locale_witness]:
+  assumes "blinding_of_on A h bo"
+  shows "blinding_of_on {x. set_list_R1 x \<subseteq> A} (hash_R1 h) (blinding_of_R1 bo)"
+  (is "blinding_of_on ?A ?h ?bo")
+proof -
+  interpret a: blinding_of_on A h bo by fact
+  show ?thesis
+  proof
+    show hash: "?bo \<le> vimage2p ?h ?h (=)" using a.hash by(rule blinding_of_hash_R1)
+
+    have "?bo x x \<and> (?bo x y \<longrightarrow> ?bo y z \<longrightarrow> ?bo x z) \<and> (?bo x y \<longrightarrow> ?bo y x \<longrightarrow> x = y)" if "x \<in> ?A" for x y z using that
+    proof(induction x arbitrary: y z)
+      case (list_R1 x y' z')
+      from list_R1.prems have s1: "set_base_F\<^sub>m x \<subseteq> A" by(fastforce)
+      from list_R1.prems have s3: "set_rec_F\<^sub>m x \<bind> set_list_R1 \<subseteq> A" by(fastforce intro: rev_bexI)
+     
+      interpret F: blinding_of_on "{y. set_base_F\<^sub>m y \<subseteq> A \<and> set_rec_F\<^sub>m y \<subseteq> set_rec_F\<^sub>m x}"
+        "hash_F h (hash_R1 h)" "blinding_of_F bo (blinding_of_R1 bo)"
+        unfolding hash_F_def blinding_of_F_def set_list_R1_eq
+      proof
+        let ?A' = "setr x \<bind> snds" and ?bo' = "rel_list_R1 bo"
+        show "?bo' x x" if "x \<in> ?A'" for x using that list_R1 by(force simp add: eq_onp_def)
+        show "?bo' x z" if "?bo' x y" "?bo' y z" "x \<in> ?A'" for x y z 
+          using that list_R1.IH[of _ x y z] list_R1.prems
+          by(force simp add: bind_UNION prod_set_defs)
+        show "x = y" if "?bo' x y" "?bo' y x" "x \<in> ?A'" for x y
+          using that list_R1.IH[of _ x y] list_R1.prems
+          by(force simp add: prod_set_defs)
+      qed(rule hash)
+      show ?case using list_R1.prems
+        apply(intro conjI)
+        subgoal using F.refl[of x] s1 unfolding blinding_of_F_def by(auto intro: list_R1.rel_intros)
+        subgoal using s1 by(auto elim!: list_R1.rel_cases F.trans[unfolded blinding_of_F_def] intro: list_R1.rel_intros)
+        subgoal using s1 by(auto elim!: list_R1.rel_cases dest: F.antisym[unfolded blinding_of_F_def])
+        done
+    qed
+    then show "x \<in> ?A \<Longrightarrow> ?bo x x" 
+      and "\<lbrakk> ?bo x y; ?bo y z; x \<in> ?A \<rbrakk> \<Longrightarrow> ?bo x z"
+      and "\<lbrakk> ?bo x y; ?bo y x; x \<in> ?A \<rbrakk> \<Longrightarrow> x = y"
+      for x y z by blast+
+  qed
+qed
+
+lemmas blinding_of_R1 [locale_witness] = blinding_of_on_R1[where A=UNIV, simplified]
+
+parametric_constant blinding_of_F_parametric[transfer_rule]: blinding_of_F_def
+
+(************************************************************)
+subsubsection \<open> Merging \<close>
+(************************************************************)
+
+definition merge_F :: "'a\<^sub>m merge \<Rightarrow> 'b\<^sub>m merge \<Rightarrow> ('a\<^sub>m, 'b\<^sub>m) list_F\<^sub>m merge" where 
+  "merge_F m mL = merge_sum merge_unit (merge_prod m mL)"
+
+lemma merge_F_cong[fundef_cong]:
+  assumes "\<And>a b. \<lbrakk> a \<in> set_base_F\<^sub>m x; b \<in> set_base_F\<^sub>m y \<rbrakk> \<Longrightarrow> m a b = m' a b"
+    and "\<And>a b. \<lbrakk> a \<in> set_rec_F\<^sub>m x; b \<in> set_rec_F\<^sub>m y \<rbrakk> \<Longrightarrow> mL a b = mL' a b"
+  shows "merge_F m mL x y = merge_F m' mL' x y"
+  using assms
+  apply(cases x; cases y)
+     apply(simp_all add: merge_F_def)
+  apply(rule arg_cong[where f="map_option _"])
+  apply(blast intro: merge_prod_cong)
+  done
+
+context
+  fixes m :: "'a\<^sub>m merge" 
+  notes setr.simps[simp]
+begin
+fun merge_R1 :: "'a\<^sub>m list_R1\<^sub>m merge" where
+  "merge_R1 (list_R1 l1) (list_R1 l2) = map_option list_R1 (merge_F m merge_R1 l1 l2)"
+end
+
+case_of_simps merge_cases [simp]: merge_R1.simps
+
+lemma merge_on_R1:
+  assumes "merge_on A h bo m"
+  shows "merge_on {x. set_list_R1 x \<subseteq> A } (hash_R1 h) (blinding_of_R1 bo) (merge_R1 m)"
+  (is "merge_on ?A ?h ?bo ?m")
+proof -
+  interpret a: merge_on A h bo m by fact
+  show ?thesis
+  proof
+    have "(?h a = ?h b \<longrightarrow> (\<exists>ab. ?m a b = Some ab \<and> ?bo a ab \<and> ?bo b ab \<and> (\<forall>u. ?bo a u \<longrightarrow> ?bo b u \<longrightarrow> ?bo ab u))) \<and>
+      (?h a \<noteq> ?h b \<longrightarrow> ?m a b = None)"
+      if "a \<in> ?A" for a b using that unfolding mem_Collect_eq
+    proof(induction a arbitrary: b rule: list_R1_induct)
+      case wfInd: (list_R1 l)
+      interpret merge_on "{y. set_base_F\<^sub>m y \<subseteq> A \<and> set_rec_F\<^sub>m y \<subseteq> set_rec_F\<^sub>m l}"
+        "hash_F h ?h" "blinding_of_F bo ?bo" "merge_F m ?m"
+        unfolding set_list_R1_eq hash_F_def merge_F_def blinding_of_F_def
+      proof
+        fix a
+        assume a: "a \<in> set_rec_F\<^sub>m l"
+        with wfInd.prems have a': "set_list_R1 a \<subseteq> A"
+          by fastforce
+
+        show "hash_R1 h a = hash_R1 h b
+           \<Longrightarrow> \<exists>ab. ?m a b = Some ab \<and> ?bo a ab \<and> ?bo b ab \<and>
+                    (\<forall>u. ?bo a u \<longrightarrow> ?bo b u \<longrightarrow> ?bo ab u)"
+          and "?h a \<noteq> ?h b \<Longrightarrow> ?m a b = None" for b      
+          using wfInd.IH[OF a a', rule_format, of b]
+          by(auto dest: sym)
+      qed
+      show ?case using wfInd.prems
+        apply(intro conjI strip)
+        subgoal
+          by(auto 4 4 dest!: join[unfolded hash_F_def]
+              simp add: blinding_of_F_def UN_subset_iff list_R1.rel_sel)
+        subgoal by(auto 4 3 intro!: undefined[simplified hash_F_def])
+        done
+    qed
+    then show
+      "?h a = ?h b \<Longrightarrow> \<exists>ab. ?m a b = Some ab \<and> ?bo a ab \<and> ?bo b ab \<and> (\<forall>u. ?bo a u \<longrightarrow> ?bo b u \<longrightarrow> ?bo ab u)"
+      "?h a \<noteq> ?h b \<Longrightarrow> ?m a b = None"
+      if "a \<in> ?A" for a b using that by blast+
+  qed
+qed
+
+lemmas merge_R1 [locale_witness] = merge_on_R1[where A=UNIV, simplified]
+
+lemma merkle_list_R1 [locale_witness]:
+  assumes "merkle_interface h bo m"
+  shows "merkle_interface (hash_R1 h) (blinding_of_R1 bo) (merge_R1 m)"
+proof -
+  interpret merge_on UNIV h bo m using assms by(unfold merkle_interface_aux)
+  show ?thesis unfolding merkle_interface_aux[symmetric] ..
+qed
+
+lemma merge_R1_cong [fundef_cong]:
+  assumes "\<And>a b. \<lbrakk> a \<in> set_list_R1 x; b \<in> set_list_R1 y \<rbrakk> \<Longrightarrow> m a b = m' a b"
+  shows "merge_R1 m x y = merge_R1 m' x y"
+  using assms
+  apply(induction x y rule: merge_R1.induct)
+  apply(simp del: merge_cases)
+  apply(rule arg_cong[where f="map_option _"])
+  apply(blast intro: merge_F_cong[unfolded bind_UNION])
+  done
+
+parametric_constant merge_F_parametric[transfer_rule]: merge_F_def
+
+lemma merge_R1_parametric [transfer_rule]:
+  includes lifting_syntax 
+  notes [simp del] = merge_cases
+  assumes [transfer_rule]: "bi_unique A"
+  shows "((A ===> A ===> rel_option A) ===> rel_list_R1 A ===> rel_list_R1 A ===> rel_option (rel_list_R1 A))
+   merge_R1 merge_R1"
+  apply(intro rel_funI)
+  subgoal premises prems [transfer_rule] for m1 m2 xs1 xs2 ys1 ys2 using prems(2, 3)
+    apply(induction xs1 ys1 arbitrary: xs2 ys2 rule: merge_R1.induct)
+    apply(elim list_R1.rel_cases rel_sum.cases; clarsimp simp add: option.rel_map merge_F_def merge_discrete_def)
+    apply(elim meta_allE; (erule meta_impE, simp)+)
+    subgoal premises [transfer_rule] by transfer_prover
+    done
+  done
+
+end
+
+subsubsection \<open> Transferring the Constructions to Lists \<close>
+type_synonym 'a\<^sub>h list\<^sub>h = "'a\<^sub>h list"
+type_synonym 'a\<^sub>m list\<^sub>m = "'a\<^sub>m list"
+
+context begin
+interpretation list_R1 .
+
+abbreviation (input) hash_list :: "('a\<^sub>m, 'a\<^sub>h) hash \<Rightarrow> ('a\<^sub>m list\<^sub>m, 'a\<^sub>h list\<^sub>h) hash"
+  where "hash_list \<equiv> map"
+abbreviation (input) blinding_of_list :: "'a\<^sub>m blinding_of \<Rightarrow> 'a\<^sub>m list\<^sub>m blinding_of"
+  where "blinding_of_list \<equiv> list_all2"
+lift_definition merge_list :: "'a\<^sub>m merge \<Rightarrow> 'a\<^sub>m list\<^sub>m merge" is merge_R1 .
+
+lemma blinding_of_list_mono:
+  "\<lbrakk> \<And>x y. bo x y \<longrightarrow> bo' x y \<rbrakk> \<Longrightarrow> 
+  blinding_of_list bo x y \<longrightarrow> blinding_of_list bo' x y"
+  by (transfer) (blast intro: list_R1.rel_mono_strong)
+
+lemmas blinding_of_list_hash = blinding_of_hash_R1[Transfer.transferred]
+  and blinding_of_on_list [locale_witness] = blinding_of_on_R1[Transfer.transferred]
+  and blinding_of_list [locale_witness] = blinding_of_R1[Transfer.transferred]
+  and merge_on_list [locale_witness] = merge_on_R1[Transfer.transferred]
+  and merge_list [locale_witness] = merge_R1[Transfer.transferred]
+  and merge_list_cong = merge_R1_cong[Transfer.transferred]
+
+lemma blinding_of_list_mono_pred:
+  "R \<le> R' \<Longrightarrow> blinding_of_list R \<le> blinding_of_list R'"
+  by(transfer) (rule list_R1.rel_mono)
+
+lemma blinding_of_list_simp: "blinding_of_list = list_all2"
+  by(transfer) (rule refl)
+
+lemma merkle_list [locale_witness]:
+  assumes [locale_witness]: "merkle_interface h bo m"
+  shows "merkle_interface (hash_list h) (blinding_of_list bo) (merge_list m)"
+  by(transfer fixing: h bo m) unfold_locales
+
+parametric_constant merge_list_parametric [transfer_rule]: merge_list_def
+
+lifting_update list.lifting
+lifting_forget list.lifting
+
+end
+
+
+(************************************************************)
+subsection \<open>Building block: function space\<close>
+(************************************************************)
+
+text \<open> We prove that we can lift the ADS construction through functions.\<close>
+
+type_synonym ('a, 'b\<^sub>h) fun\<^sub>h = "'a \<Rightarrow> 'b\<^sub>h"
+type_notation fun\<^sub>h (infixr "\<Rightarrow>\<^sub>h" 0)
+
+type_synonym ('a, 'b\<^sub>m) fun\<^sub>m = "'a \<Rightarrow> 'b\<^sub>m"
+type_notation fun\<^sub>m (infixr "\<Rightarrow>\<^sub>m" 0)
+
+(************************************************************)
+subsubsection \<open> Hashes \<close>
+(************************************************************)
+
+text \<open> Only the range is live, the domain is dead like for BNFs. \<close>
+
+abbreviation (input) hash_fun' :: "('a \<Rightarrow>\<^sub>m 'b\<^sub>h, 'a \<Rightarrow>\<^sub>h 'b\<^sub>h) hash" where
+  "hash_fun' \<equiv> id"
+
+abbreviation (input) hash_fun :: "('b\<^sub>m, 'b\<^sub>h) hash \<Rightarrow> ('a \<Rightarrow>\<^sub>m 'b\<^sub>m, 'a \<Rightarrow>\<^sub>h 'b\<^sub>h) hash"
+  where "hash_fun \<equiv> comp"
+
+(************************************************************)
+subsubsection \<open> Blinding \<close>
+(************************************************************)
+
+abbreviation (input) blinding_of_fun :: "'b\<^sub>m blinding_of \<Rightarrow> ('a \<Rightarrow>\<^sub>m 'b\<^sub>m) blinding_of" where
+  "blinding_of_fun \<equiv> rel_fun (=)"
+
+lemmas blinding_of_fun_mono = fun.rel_mono
+
+lemma blinding_of_fun_hash:
+  assumes "bo \<le> vimage2p rh rh (=)"
+  shows "blinding_of_fun bo \<le> vimage2p (hash_fun rh) (hash_fun rh) (=)"
+  using assms by(auto simp add: vimage2p_def rel_fun_def le_fun_def)
+
+lemma blinding_of_on_fun [locale_witness]:
+  assumes "blinding_of_on A rh bo"
+  shows "blinding_of_on {x. range x \<subseteq> A} (hash_fun rh) (blinding_of_fun bo)"
+  (is "blinding_of_on ?A ?h ?bo")
+proof -
+  interpret a: blinding_of_on A rh bo by fact
+  show ?thesis
+  proof
+    show "?bo x x" if "x \<in> ?A" for x using that by(auto simp add: rel_fun_def intro: a.refl)
+    show "?bo x z" if "?bo x y" "?bo y z" "x \<in> ?A" for x y z using that
+      by(auto 4 3 simp add: rel_fun_def intro: a.trans)
+    show "x = y" if "?bo x y" "?bo y x" "x \<in> ?A" for x y using that
+      by(fastforce simp add: fun_eq_iff rel_fun_def intro: a.antisym)
+  qed(rule blinding_of_fun_hash a.hash)+
+qed
+
+lemmas blinding_of_fun [locale_witness] = blinding_of_on_fun[where A=UNIV, simplified]
+
+(************************************************************)
+subsubsection \<open> Merging \<close>
+(************************************************************)
+
+context
+  fixes m :: "'b\<^sub>m merge"
+begin
+
+definition merge_fun :: "('a \<Rightarrow>\<^sub>m 'b\<^sub>m) merge" where
+  "merge_fun f g = (if \<forall>x. m (f x) (g x) \<noteq> None then Some (\<lambda>x. the (m (f x) (g x))) else None)"
+
+lemma merge_on_fun [locale_witness]:
+  assumes "merge_on A rh bo m"
+  shows "merge_on {x. range x \<subseteq> A} (hash_fun rh) (blinding_of_fun bo) merge_fun"
+  (is "merge_on ?A ?h ?bo ?m")
+proof -
+  interpret a: merge_on A rh bo m by fact
+  show ?thesis
+  proof
+    show "\<exists>ab. ?m a b = Some ab \<and> ?bo a ab \<and> ?bo b ab \<and> (\<forall>u. ?bo a u \<longrightarrow> ?bo b u \<longrightarrow> ?bo ab u)"
+      if "?h a = ?h b" "a \<in> ?A" for a b 
+      using that(1)[THEN fun_cong, unfolded o_apply, THEN a.join, OF that(2)[unfolded mem_Collect_eq, THEN subsetD, OF rangeI]]
+      by atomize(subst (asm) choice_iff; auto simp add: merge_fun_def rel_fun_def)
+    show "?m a b = None" if "?h a \<noteq> ?h b" "a \<in> ?A" for a b using that
+      by(auto simp add: merge_fun_def fun_eq_iff dest: a.undefined)
+  qed
+qed
+
+lemmas merge_fun [locale_witness] = merge_on_fun[where A=UNIV, simplified]
+
+end
+
+lemma merge_fun_cong[fundef_cong]:
+  assumes "\<And>a b. \<lbrakk> a \<in> range f; b \<in> range g \<rbrakk> \<Longrightarrow> m a b = m' a b"
+  shows "merge_fun m f g = merge_fun m' f g"
+  using assms[OF rangeI rangeI] by(clarsimp simp add: merge_fun_def)
+
+lemma is_none_alt_def: "Option.is_none x \<longleftrightarrow> (case x of None \<Rightarrow> True | Some _ \<Rightarrow> False)"
+  by(auto simp add: Option.is_none_def split: option.splits)
+
+parametric_constant is_none_parametric [transfer_rule]: is_none_alt_def
+
+lemma merge_fun_parametric [transfer_rule]: includes lifting_syntax shows
+  "((A ===> B ===> rel_option C) ===> ((=) ===> A) ===> ((=) ===> B) ===> rel_option ((=) ===> C))
+   merge_fun merge_fun"
+proof(intro rel_funI)
+  fix m :: "'a merge" and m' :: "'b merge" and f :: "'c \<Rightarrow> 'a" and f' :: "'c \<Rightarrow> 'b" and g :: "'c \<Rightarrow> 'a" and g' :: "'c \<Rightarrow> 'b"
+  assume m: "(A ===> B ===> rel_option C) m m'"
+    and f: "((=) ===> A) f f'" and g: "((=) ===> B) g g'"
+  note [transfer_rule] = this
+  have cond [unfolded Option.is_none_def]: "(\<forall>x. \<not> Option.is_none (m (f x) (g x))) \<longleftrightarrow> (\<forall>x. \<not> Option.is_none (m' (f' x) (g' x)))" 
+    by transfer_prover
+  moreover 
+  have "((=) ===> C) (\<lambda>x. the (m (f x) (g x))) (\<lambda>x. the (m' (f' x) (g' x)))" if *: "\<forall>x. \<not> m (f x) (g x) = None"
+  proof -
+    obtain fg fg' where m: "m (f x) (g x) = Some (fg x)" and m': "m' (f' x) (g' x) = Some (fg' x)" for x
+      using * *[simplified cond]
+      by(simp)(subst (asm) (1 2) choice_iff; clarsimp)
+    have "rel_option C (Some (fg x)) (Some (fg' x))" for x unfolding m[symmetric] m'[symmetric] by transfer_prover
+    then show ?thesis by(simp add: rel_fun_def m m')
+  qed
+  ultimately show "rel_option ((=) ===> C) (merge_fun m f g) (merge_fun m' f' g')"
+    unfolding merge_fun_def by(simp)
+qed
+
+(************************************************************)
+subsubsection \<open> Merkle Interface \<close>
+(************************************************************)
+
+lemma merkle_fun [locale_witness]:
+  assumes "merkle_interface rh bo m"
+  shows "merkle_interface (hash_fun rh) (blinding_of_fun bo) (merge_fun m)"
+proof -
+  interpret a: merge_on UNIV rh bo m unfolding merkle_interface_aux[symmetric] by fact
+  show ?thesis unfolding merkle_interface_aux[symmetric] ..
+qed
+
+(************************************************************)
+subsection \<open>Rose trees\<close>
+(************************************************************)
+
+text \<open> 
+We now define an ADS over rose trees, which is like a arbitrarily branching Merkle tree where each
+node in the tree can be blinded, including the root. The number of children and the position of a
+child among its siblings cannot be hidden. The construction allows to plug in further blindable
+positions in the labels of the nodes.
+\<close>
+
+type_synonym ('a, 'b) rose_tree_F = "'a \<times> 'b list"
+
+abbreviation (input) map_rose_tree_F where
+  "map_rose_tree_F f1 f2 \<equiv> map_prod f1 (map f2)"
+definition map_rose_tree_F_const where
+  "map_rose_tree_F_const f1 f2 \<equiv> map_rose_tree_F f1 f2"
+
+datatype 'a rose_tree = Tree "('a, 'a rose_tree) rose_tree_F"
+
+type_synonym ('a\<^sub>h, 'b\<^sub>h) rose_tree_F\<^sub>h = "('a\<^sub>h \<times>\<^sub>h 'b\<^sub>h list\<^sub>h) blindable\<^sub>h"
+
+datatype 'a\<^sub>h rose_tree\<^sub>h = Tree\<^sub>h "('a\<^sub>h, 'a\<^sub>h rose_tree\<^sub>h) rose_tree_F\<^sub>h"
+
+type_synonym ('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) rose_tree_F\<^sub>m = "('a\<^sub>m \<times>\<^sub>m 'b\<^sub>m list\<^sub>m, 'a\<^sub>h \<times>\<^sub>h 'b\<^sub>h list\<^sub>h) blindable\<^sub>m"
+
+datatype ('a\<^sub>m, 'a\<^sub>h) rose_tree\<^sub>m = Tree\<^sub>m "('a\<^sub>m, 'a\<^sub>h, ('a\<^sub>m, 'a\<^sub>h) rose_tree\<^sub>m, 'a\<^sub>h rose_tree\<^sub>h) rose_tree_F\<^sub>m"
+
+abbreviation (input) map_rose_tree_F\<^sub>m
+  :: "('ma \<Rightarrow> 'a) \<Rightarrow> ('mr \<Rightarrow> 'r) \<Rightarrow> ('ma, 'ha, 'mr, 'hr) rose_tree_F\<^sub>m \<Rightarrow> ('a, 'ha, 'r, 'hr) rose_tree_F\<^sub>m"
+  where
+  "map_rose_tree_F\<^sub>m f g \<equiv> map_blindable\<^sub>m (map_prod f (map g)) id"
+
+(************************************************************)
+subsubsection \<open> Hashes \<close>
+(************************************************************)
+
+abbreviation (input) hash_rt_F' 
+  :: "(('a\<^sub>h, 'a\<^sub>h, 'b\<^sub>h, 'b\<^sub>h) rose_tree_F\<^sub>m, ('a\<^sub>h, 'b\<^sub>h) rose_tree_F\<^sub>h) hash"
+  where
+  "hash_rt_F' \<equiv> hash_blindable id"
+
+definition hash_rt_F\<^sub>m
+  :: "('a\<^sub>m, 'a\<^sub>h) hash \<Rightarrow> ('b\<^sub>m, 'b\<^sub>h) hash \<Rightarrow> 
+    (('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) rose_tree_F\<^sub>m, ('a\<^sub>h, 'b\<^sub>h) rose_tree_F\<^sub>h) hash" where
+  "hash_rt_F\<^sub>m h rhm \<equiv> hash_rt_F' o map_rose_tree_F\<^sub>m h rhm"
+
+lemma hash_rt_F\<^sub>m_alt_def: "hash_rt_F\<^sub>m h rhm = hash_blindable (map_prod h (map rhm))"
+  by(simp add: hash_rt_F\<^sub>m_def fun_eq_iff hash_map_blindable_simp)
+
+primrec (transfer) hash_rt_tree'
+  :: "(('a\<^sub>h, 'a\<^sub>h) rose_tree\<^sub>m, 'a\<^sub>h rose_tree\<^sub>h) hash" where
+  "hash_rt_tree' (Tree\<^sub>m x) = Tree\<^sub>h (hash_rt_F' (map_rose_tree_F\<^sub>m id hash_rt_tree' x))"
+
+definition hash_tree
+  :: "('a\<^sub>m, 'a\<^sub>h) hash \<Rightarrow> (('a\<^sub>m, 'a\<^sub>h) rose_tree\<^sub>m, 'a\<^sub>h rose_tree\<^sub>h) hash" where
+  "hash_tree h = hash_rt_tree' o map_rose_tree\<^sub>m h id"
+
+lemma blindable\<^sub>m_map_compositionality:
+  "map_blindable\<^sub>m f g o map_blindable\<^sub>m f' g' = map_blindable\<^sub>m (f o f') (g o g')"
+  by(rule ext) (simp add: blindable\<^sub>m.map_comp)
+
+lemma hash_tree_simps [simp]:
+  "hash_tree h (Tree\<^sub>m x) = Tree\<^sub>h (hash_rt_F\<^sub>m h (hash_tree h) x)"
+  by(simp add: hash_tree_def hash_rt_F\<^sub>m_def 
+      map_prod.comp map_sum.comp rose_tree\<^sub>h.map_comp blindable\<^sub>m.map_comp
+      prod.map_id0 rose_tree\<^sub>h.map_id0)
+
+parametric_constant hash_rt_F\<^sub>m_parametric [transfer_rule]: hash_rt_F\<^sub>m_alt_def
+
+parametric_constant hash_tree_parametric [transfer_rule]: hash_tree_def
+
+(************************************************************)
+subsubsection \<open> Blinding \<close>
+(************************************************************)
+
+abbreviation (input) blinding_of_rt_F\<^sub>m
+  :: "('a\<^sub>m, 'a\<^sub>h) hash \<Rightarrow> 'a\<^sub>m blinding_of \<Rightarrow> ('b\<^sub>m, 'b\<^sub>h) hash \<Rightarrow> 'b\<^sub>m blinding_of
+      \<Rightarrow> ('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) rose_tree_F\<^sub>m blinding_of" where
+  "blinding_of_rt_F\<^sub>m ha boa hb bob \<equiv> blinding_of_blindable (hash_prod ha (map hb))
+    (blinding_of_prod boa (blinding_of_list bob))"
+
+lemma blinding_of_rt_F\<^sub>m_mono:
+  "\<lbrakk> boa \<le> boa'; bob \<le> bob' \<rbrakk> \<Longrightarrow> blinding_of_rt_F\<^sub>m ha boa hb bob \<le> blinding_of_rt_F\<^sub>m ha boa' hb bob'"
+  by(intro blinding_of_blindable_mono prod.rel_mono list.rel_mono)
+
+lemma blinding_of_rt_F\<^sub>m_mono_inductive:
+  assumes "\<And>x y. boa x y \<longrightarrow> boa' x y" "\<And>x y. bob x y \<longrightarrow> bob' x y"
+  shows "blinding_of_rt_F\<^sub>m ha boa hb bob x y \<longrightarrow> blinding_of_rt_F\<^sub>m ha boa' hb bob' x y"
+  apply(rule impI)
+  apply(erule blinding_of_rt_F\<^sub>m_mono[THEN predicate2D, rotated -1])
+  using assms by blast+
+
+context
+  fixes h :: "('a\<^sub>m, 'a\<^sub>h) hash" 
+    and bo :: "'a\<^sub>m blinding_of"
+begin
+
+inductive blinding_of_tree :: "('a\<^sub>m, 'a\<^sub>h) rose_tree\<^sub>m blinding_of" where
+  "blinding_of_tree (Tree\<^sub>m t1) (Tree\<^sub>m t2)" 
+  if "blinding_of_rt_F\<^sub>m h bo (hash_tree h) blinding_of_tree t1 t2"
+monos blinding_of_rt_F\<^sub>m_mono_inductive
+
+end
+
+inductive_simps blinding_of_tree_simps [simp]:
+  "blinding_of_tree h bo (Tree\<^sub>m t1) (Tree\<^sub>m t2)"
+
+lemma blinding_of_rt_F\<^sub>m_hash:
+  assumes "boa \<le> vimage2p ha ha (=)" "bob \<le> vimage2p hb hb (=)"
+  shows "blinding_of_rt_F\<^sub>m ha boa hb bob \<le> vimage2p (hash_rt_F\<^sub>m ha hb) (hash_rt_F\<^sub>m ha hb) (=)"
+  apply(rule order_trans)
+   apply(rule blinding_of_blindable_hash)
+   apply(fold relator_eq)
+   apply(unfold vimage2p_map_rel_prod vimage2p_map_list_all2)
+   apply(rule prod.rel_mono assms list.rel_mono)+
+  apply(simp only: hash_rt_F\<^sub>m_def vimage2p_comp o_apply hash_blindable_def blindable\<^sub>m.map_id0 id_def[symmetric] vimage2p_id id_apply)
+  done
+
+lemma blinding_of_tree_hash:
+  assumes "bo \<le> vimage2p h h (=)"
+  shows "blinding_of_tree h bo \<le> vimage2p (hash_tree h) (hash_tree h) (=)"
+  apply(rule predicate2I vimage2pI)+
+  apply(erule blinding_of_tree.induct)
+  apply(simp)
+  apply(erule blinding_of_rt_F\<^sub>m_hash[OF assms, THEN predicate2D_vimage2p, rotated 1])
+  apply(blast intro: vimage2pI)
+  done
+
+abbreviation (input) set1_rt_F\<^sub>m :: "('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>h, 'b\<^sub>m) rose_tree_F\<^sub>m \<Rightarrow> 'a\<^sub>m set" where
+  "set1_rt_F\<^sub>m x \<equiv> set1_blindable\<^sub>m x \<bind> fsts"
+
+abbreviation (input) set3_rt_F\<^sub>m :: "('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) rose_tree_F\<^sub>m \<Rightarrow> 'b\<^sub>m set" where 
+  "set3_rt_F\<^sub>m x \<equiv> (set1_blindable\<^sub>m x \<bind> snds) \<bind> set"
+
+lemma set_rt_F\<^sub>m_eq: 
+  "{x. set1_rt_F\<^sub>m x \<subseteq> A \<and> set3_rt_F\<^sub>m x \<subseteq> B} = 
+   {x. set1_blindable\<^sub>m x \<subseteq> {x. fsts x \<subseteq> A \<and> snds x \<subseteq> {x. set x \<subseteq> B}}}"
+  by force
+
+lemma hash_blindable_map: "hash_blindable f \<circ> map_blindable\<^sub>m g id = hash_blindable (f \<circ> g)" 
+  by(rule ext) (simp add: hash_blindable_def blindable\<^sub>m.map_comp)
+
+lemma blinding_of_on_tree [locale_witness]:
+  assumes "blinding_of_on A h bo"
+  shows "blinding_of_on {x. set1_rose_tree\<^sub>m x \<subseteq> A} (hash_tree h) (blinding_of_tree h bo)"
+  (is "blinding_of_on ?A ?h ?bo")
+proof -
+  interpret a: blinding_of_on A h bo by fact
+  show ?thesis
+  proof
+    show "?bo \<le> vimage2p ?h ?h (=)" using a.hash by(rule blinding_of_tree_hash)
+    have "?bo x x \<and> (?bo x y \<longrightarrow> ?bo y z \<longrightarrow> ?bo x z) \<and> (?bo x y \<longrightarrow> ?bo y x \<longrightarrow> x = y)" if "x \<in> ?A" for x y z using that
+    proof(induction x arbitrary: y z)
+      case (Tree\<^sub>m x)
+      have [locale_witness]: "blinding_of_on (set3_rt_F\<^sub>m x) (hash_tree h) (blinding_of_tree h bo)"
+        apply unfold_locales
+        subgoal by(rule blinding_of_tree_hash)(rule a.hash)
+        subgoal using Tree\<^sub>m.IH Tree\<^sub>m.prems by(fastforce simp add: eq_onp_def)
+        subgoal for x y z using Tree\<^sub>m.IH[of _ _ x y z] Tree\<^sub>m.prems by fastforce
+        subgoal for x y using Tree\<^sub>m.IH[of _ _ x y] Tree\<^sub>m.prems by fastforce
+        done
+      interpret blinding_of_on
+        "{a. set1_rt_F\<^sub>m a \<subseteq> A \<and> set3_rt_F\<^sub>m a \<subseteq> set3_rt_F\<^sub>m x}"
+        "hash_rt_F\<^sub>m h ?h" "blinding_of_rt_F\<^sub>m h bo ?h ?bo" 
+        unfolding set_rt_F\<^sub>m_eq hash_rt_F\<^sub>m_alt_def ..
+      from Tree\<^sub>m.prems show ?case
+        apply(intro conjI)
+        subgoal by(fastforce intro!: blinding_of_tree.intros refl[unfolded hash_rt_F\<^sub>m_alt_def])
+        subgoal by(fastforce elim!: blinding_of_tree.cases trans[unfolded hash_rt_F\<^sub>m_alt_def] 
+                    intro!: blinding_of_tree.intros)
+        subgoal by(fastforce elim!: blinding_of_tree.cases antisym[unfolded hash_rt_F\<^sub>m_alt_def])
+        done
+    qed
+    then show "x \<in> ?A \<Longrightarrow> ?bo x x"
+      and "\<lbrakk> ?bo x y; ?bo y z; x \<in> ?A \<rbrakk> \<Longrightarrow> ?bo x z"
+      and "\<lbrakk> ?bo x y; ?bo y x; x \<in> ?A \<rbrakk> \<Longrightarrow> x = y"
+      for x y z by blast+
+  qed
+qed
+
+lemmas blinding_of_tree [locale_witness] = blinding_of_on_tree[where A=UNIV, simplified]
+
+lemma blinding_of_tree_mono:
+  "bo \<le> bo' \<Longrightarrow> blinding_of_tree h bo \<le> blinding_of_tree h bo'"
+  apply(rule predicate2I)
+  apply(erule blinding_of_tree.induct)
+  apply(rule blinding_of_tree.intros)
+  apply(erule blinding_of_rt_F\<^sub>m_mono[THEN predicate2D, rotated -1])
+  apply(blast)+
+  done
+
+(************************************************************)
+subsubsection \<open> Merging \<close>
+(************************************************************)
+
+definition merge_rt_F\<^sub>m 
+  :: "('a\<^sub>m, 'a\<^sub>h) hash \<Rightarrow> 'a\<^sub>m merge \<Rightarrow> ('b\<^sub>m, 'b\<^sub>h) hash \<Rightarrow> 'b\<^sub>m merge \<Rightarrow>
+      ('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) rose_tree_F\<^sub>m merge"
+  where
+  "merge_rt_F\<^sub>m ha ma hr mr \<equiv> merge_blindable (hash_prod ha (hash_list hr)) (merge_prod ma (merge_list mr))"
+
+lemma merge_rt_F\<^sub>m_cong [fundef_cong]:
+  assumes "\<And>a b. \<lbrakk> a \<in> set1_rt_F\<^sub>m x; b \<in> set1_rt_F\<^sub>m y \<rbrakk> \<Longrightarrow> ma a b = ma' a b"
+    and "\<And>a b. \<lbrakk> a \<in> set3_rt_F\<^sub>m x; b \<in> set3_rt_F\<^sub>m y \<rbrakk> \<Longrightarrow> mm a b = mm' a b"
+  shows "merge_rt_F\<^sub>m ha ma hm mm x y = merge_rt_F\<^sub>m ha ma' hm mm' x y"
+  using assms
+  apply(cases x; cases y; simp add: merge_rt_F\<^sub>m_def bind_UNION)
+  apply(rule arg_cong[where f="map_option _"])
+  apply(blast intro: merge_prod_cong merge_list_cong)
+  done
+
+lemma in_set1_blindable\<^sub>m_iff: "x \<in> set1_blindable\<^sub>m y \<longleftrightarrow> y = Unblinded x"
+  by(cases y) auto
+
+context 
+  fixes h :: "('a\<^sub>m, 'a\<^sub>h) hash"
+    and ma :: "'a\<^sub>m merge"
+  notes in_set1_blindable\<^sub>m_iff[simp]
+begin
+fun merge_tree :: "('a\<^sub>m, 'a\<^sub>h) rose_tree\<^sub>m merge" where
+  "merge_tree (Tree\<^sub>m x) (Tree\<^sub>m y) = map_option Tree\<^sub>m (
+    merge_rt_F\<^sub>m h ma (hash_tree h) merge_tree x y)"
+end
+
+lemma merge_on_tree [locale_witness]:
+  assumes "merge_on A h bo m"
+  shows "merge_on {x. set1_rose_tree\<^sub>m x \<subseteq> A} (hash_tree h) (blinding_of_tree h bo) (merge_tree h m)"
+  (is "merge_on ?A ?h ?bo ?m")
+proof -
+  interpret a: merge_on A h bo m by fact
+  show ?thesis 
+  proof
+    have "(?h a = ?h b \<longrightarrow> (\<exists>ab. ?m a b = Some ab \<and> ?bo a ab \<and> ?bo b ab \<and> (\<forall>u. ?bo a u \<longrightarrow> ?bo b u \<longrightarrow> ?bo ab u))) \<and>
+      (?h a \<noteq> ?h b \<longrightarrow> ?m a b = None)"
+      if "a \<in> ?A" for a b using that unfolding mem_Collect_eq
+    proof(induction a arbitrary: b rule: rose_tree\<^sub>m.induct)
+      case (Tree\<^sub>m x y)
+      interpret merge_on 
+        "{y. set1_rt_F\<^sub>m y \<subseteq> A \<and> set3_rt_F\<^sub>m y \<subseteq> set3_rt_F\<^sub>m x}"
+        "hash_rt_F\<^sub>m h ?h"
+        "blinding_of_rt_F\<^sub>m h bo ?h ?bo"
+        "merge_rt_F\<^sub>m h m ?h ?m"
+        unfolding set_rt_F\<^sub>m_eq hash_rt_F\<^sub>m_alt_def merge_rt_F\<^sub>m_def
+      proof
+        fix a
+        assume a: "a \<in> set3_rt_F\<^sub>m x"
+        with Tree\<^sub>m.prems have a': "set1_rose_tree\<^sub>m a \<subseteq> A"
+          by(force simp add: bind_UNION)
+
+        from a obtain l and ab where a'': "ab \<in> set1_blindable\<^sub>m x" "l \<in> snds ab" "a \<in> set l" 
+          by(clarsimp simp add: bind_UNION)
+
+        fix b
+        from Tree\<^sub>m.IH[OF a'' a', rule_format, of b]
+        show "hash_tree h a = hash_tree h b
+           \<Longrightarrow> \<exists>ab. merge_tree h m a b = Some ab \<and> blinding_of_tree h bo a ab \<and> blinding_of_tree h bo b ab \<and>
+                    (\<forall>u. blinding_of_tree h bo a u \<longrightarrow> blinding_of_tree h bo b u \<longrightarrow> blinding_of_tree h bo ab u)"
+          and "hash_tree h a \<noteq> hash_tree h b \<Longrightarrow> merge_tree h m a b = None"
+          by(auto dest: sym)
+      qed
+      show ?case using Tree\<^sub>m.prems
+        apply(intro conjI strip)
+        subgoal by(cases y)(fastforce dest!: join simp add: blinding_of_tree.simps)
+        subgoal by (cases y) (fastforce dest!: undefined)
+        done
+    qed
+    then show
+      "?h a = ?h b \<Longrightarrow> \<exists>ab. ?m a b = Some ab \<and> ?bo a ab \<and> ?bo b ab \<and> (\<forall>u. ?bo a u \<longrightarrow> ?bo b u \<longrightarrow> ?bo ab u)"
+      "?h a \<noteq> ?h b \<Longrightarrow> ?m a b = None"
+      if "a \<in> ?A" for a b using that by blast+
+  qed
+qed
+
+lemmas merge_tree [locale_witness] = merge_on_tree[where A=UNIV, simplified]
+
+lemma option_bind_comm:
+ "((x :: 'a option) \<bind> (\<lambda>y. c \<bind> (\<lambda>z. f y z))) = (c \<bind> (\<lambda>y. x \<bind> (\<lambda>z. f z y)))"
+  by(cases x; cases c; auto)
+
+parametric_constant merge_rt_F\<^sub>m_parametric [transfer_rule]: merge_rt_F\<^sub>m_def
+
+(************************************************************) 
+subsubsection \<open>Merkle interface\<close>
+(************************************************************)
+
+lemma merkle_tree [locale_witness]:
+  assumes "merkle_interface h bo m"
+  shows "merkle_interface (hash_tree h) (blinding_of_tree h bo) (merge_tree h m)"
+proof -
+  interpret merge_on UNIV h bo m using assms unfolding merkle_interface_aux .
+  show ?thesis unfolding merkle_interface_aux[symmetric] ..
+qed
+
+lemma merge_tree_cong [fundef_cong]:
+  assumes "\<And>a b. \<lbrakk> a \<in> set1_rose_tree\<^sub>m x; b \<in> set1_rose_tree\<^sub>m y \<rbrakk> \<Longrightarrow> m a b = m' a b"
+  shows "merge_tree h m x y = merge_tree h m' x y"
+  using assms
+  apply(induction x y rule: merge_tree.induct)
+  apply(simp add: bind_UNION)
+  apply(rule arg_cong[where f="map_option _"])
+  apply(rule merge_rt_F\<^sub>m_cong; simp add: bind_UNION; blast)
+  done
+
+end

formal/afp/ADS_Functor/Canton_Transaction_Tree.thy

@@ -0,0 +1,518 @@
+theory Canton_Transaction_Tree imports
+  Inclusion_Proof_Construction
+begin
+
+section \<open>Canton's hierarchical transaction trees\<close>
+
+typedecl view_data
+typedecl view_metadata
+typedecl common_metadata
+typedecl participant_metadata
+
+datatype view = View view_metadata view_data (subviews: "view list")
+
+datatype transaction = Transaction common_metadata participant_metadata (views: "view list")
+
+subsection \<open>Views as authenticated data structures\<close>
+
+type_synonym view_metadata\<^sub>h = "view_metadata blindable\<^sub>h"
+type_synonym view_data\<^sub>h = "view_data blindable\<^sub>h"
+
+datatype view\<^sub>h = View\<^sub>h "((view_metadata\<^sub>h \<times>\<^sub>h view_data\<^sub>h) \<times>\<^sub>h view\<^sub>h list\<^sub>h) blindable\<^sub>h"
+
+type_synonym view_metadata\<^sub>m = "(view_metadata, view_metadata) blindable\<^sub>m"
+type_synonym view_data\<^sub>m = "(view_data, view_data) blindable\<^sub>m"
+
+datatype view\<^sub>m = View\<^sub>m 
+  "((view_metadata\<^sub>m \<times>\<^sub>m view_data\<^sub>m) \<times>\<^sub>m view\<^sub>m list\<^sub>m,
+    (view_metadata\<^sub>h \<times>\<^sub>h view_data\<^sub>h) \<times>\<^sub>h view\<^sub>h list\<^sub>h) blindable\<^sub>m"
+
+abbreviation (input) hash_view_data :: "(view_data\<^sub>m, view_data\<^sub>h) hash" where
+  "hash_view_data \<equiv> hash_blindable id"
+abbreviation (input) blinding_of_view_data :: "view_data\<^sub>m blinding_of" where
+  "blinding_of_view_data \<equiv> blinding_of_blindable id (=)"
+abbreviation (input) merge_view_data :: "view_data\<^sub>m merge" where
+  "merge_view_data \<equiv> merge_blindable id merge_discrete"
+
+lemma merkle_view_data:
+  "merkle_interface hash_view_data blinding_of_view_data merge_view_data"
+  by unfold_locales
+
+abbreviation (input) hash_view_metadata :: "(view_metadata\<^sub>m, view_metadata\<^sub>h) hash" where
+  "hash_view_metadata \<equiv> hash_blindable id"
+abbreviation (input) blinding_of_view_metadata :: "view_metadata\<^sub>m blinding_of" where
+  "blinding_of_view_metadata \<equiv> blinding_of_blindable id (=)"
+abbreviation (input) merge_view_metadata :: "view_metadata\<^sub>m merge" where
+  "merge_view_metadata \<equiv> merge_blindable id merge_discrete"
+
+lemma merkle_view_metadata:
+  "merkle_interface hash_view_metadata blinding_of_view_metadata merge_view_metadata"
+  by unfold_locales
+
+type_synonym view_content = "view_metadata \<times> view_data"
+type_synonym view_content\<^sub>h = "view_metadata\<^sub>h \<times>\<^sub>h view_data\<^sub>h"
+type_synonym view_content\<^sub>m = "view_metadata\<^sub>m \<times>\<^sub>m view_data\<^sub>m"
+
+locale view_merkle begin
+
+type_synonym view\<^sub>h' = "view_content\<^sub>h rose_tree\<^sub>h"
+
+primrec from_view\<^sub>h :: "view\<^sub>h \<Rightarrow> view\<^sub>h'" where
+  "from_view\<^sub>h (View\<^sub>h x) = Tree\<^sub>h (map_blindable\<^sub>h (map_prod id (map from_view\<^sub>h)) x)"
+
+primrec to_view\<^sub>h :: "view\<^sub>h' \<Rightarrow> view\<^sub>h" where
+  "to_view\<^sub>h (Tree\<^sub>h x) = View\<^sub>h (map_blindable\<^sub>h (map_prod id (map to_view\<^sub>h)) x)"
+
+lemma from_to_view\<^sub>h [simp]: "from_view\<^sub>h (to_view\<^sub>h x) = x"
+  apply(induction x)
+  apply(simp add: blindable\<^sub>h.map_comp o_def prod.map_comp)
+  apply(simp cong: blindable\<^sub>h.map_cong prod.map_cong list.map_cong add: blindable\<^sub>h.map_id[unfolded id_def])
+  done
+
+lemma to_from_view\<^sub>h [simp]: "to_view\<^sub>h (from_view\<^sub>h x) = x"
+  apply(induction x)
+  apply(simp add: blindable\<^sub>h.map_comp o_def prod.map_comp)
+  apply(simp cong: blindable\<^sub>h.map_cong prod.map_cong list.map_cong add: blindable\<^sub>h.map_id[unfolded id_def])
+  done
+
+lemma iso_view\<^sub>h: "type_definition from_view\<^sub>h to_view\<^sub>h UNIV"
+  by unfold_locales simp_all
+
+setup_lifting iso_view\<^sub>h
+
+lemma cr_view\<^sub>h_Grp: "cr_view\<^sub>h = Grp UNIV to_view\<^sub>h"
+  by(simp add: cr_view\<^sub>h_def Grp_def fun_eq_iff)(transfer, auto)
+
+lemma View\<^sub>h_transfer [transfer_rule]: includes lifting_syntax shows
+  "(rel_blindable\<^sub>h (rel_prod (=) (list_all2 pcr_view\<^sub>h)) ===> pcr_view\<^sub>h) Tree\<^sub>h View\<^sub>h"
+  by(simp add: rel_fun_def view\<^sub>h.pcr_cr_eq cr_view\<^sub>h_Grp list.rel_Grp eq_alt prod.rel_Grp blindable\<^sub>h.rel_Grp)
+    (simp add: Grp_def)
+
+type_synonym view\<^sub>m' = "(view_content\<^sub>m, view_content\<^sub>h) rose_tree\<^sub>m"
+
+primrec from_view\<^sub>m :: "view\<^sub>m \<Rightarrow> view\<^sub>m'" where
+  "from_view\<^sub>m (View\<^sub>m x) = Tree\<^sub>m (map_blindable\<^sub>m (map_prod id (map from_view\<^sub>m)) (map_prod id (map from_view\<^sub>h)) x)"
+
+primrec to_view\<^sub>m :: "view\<^sub>m' \<Rightarrow> view\<^sub>m" where
+  "to_view\<^sub>m (Tree\<^sub>m x) = View\<^sub>m (map_blindable\<^sub>m (map_prod id (map to_view\<^sub>m)) (map_prod id (map to_view\<^sub>h)) x)"
+
+lemma from_to_view\<^sub>m [simp]: "from_view\<^sub>m (to_view\<^sub>m x) = x"
+  apply(induction x)
+  apply(simp add: blindable\<^sub>m.map_comp o_def prod.map_comp)
+  apply(simp cong: blindable\<^sub>m.map_cong prod.map_cong list.map_cong add: blindable\<^sub>m.map_id[unfolded id_def])
+  done
+
+lemma to_from_view\<^sub>m [simp]: "to_view\<^sub>m (from_view\<^sub>m x) = x"
+  apply(induction x)
+  apply(simp add: blindable\<^sub>m.map_comp o_def prod.map_comp)
+  apply(simp cong: blindable\<^sub>m.map_cong prod.map_cong list.map_cong add: blindable\<^sub>m.map_id[unfolded id_def])
+  done
+
+lemma iso_view\<^sub>m: "type_definition from_view\<^sub>m to_view\<^sub>m UNIV"
+  by unfold_locales simp_all
+
+setup_lifting iso_view\<^sub>m
+
+lemma cr_view\<^sub>m_Grp: "cr_view\<^sub>m = Grp UNIV to_view\<^sub>m"
+  by(simp add: cr_view\<^sub>m_def Grp_def fun_eq_iff)(transfer, auto)
+
+lemma View\<^sub>m_transfer [transfer_rule]: includes lifting_syntax shows
+  "(rel_blindable\<^sub>m (rel_prod (=) (list_all2 pcr_view\<^sub>m)) (rel_prod (=) (list_all2 pcr_view\<^sub>h)) ===> pcr_view\<^sub>m) Tree\<^sub>m View\<^sub>m"
+  by(simp add: rel_fun_def view\<^sub>h.pcr_cr_eq view\<^sub>m.pcr_cr_eq cr_view\<^sub>h_Grp cr_view\<^sub>m_Grp list.rel_Grp eq_alt prod.rel_Grp blindable\<^sub>m.rel_Grp)
+    (simp add: Grp_def)
+
+end
+
+code_datatype View\<^sub>h
+code_datatype View\<^sub>m
+
+context begin
+interpretation view_merkle .
+
+abbreviation (input) hash_view_content :: "(view_content\<^sub>m, view_content\<^sub>h) hash" where
+  "hash_view_content \<equiv> hash_prod hash_view_metadata hash_view_data"
+
+abbreviation (input) blinding_of_view_content :: "view_content\<^sub>m blinding_of" where
+  "blinding_of_view_content \<equiv> blinding_of_prod blinding_of_view_metadata blinding_of_view_data"
+
+abbreviation (input) merge_view_content :: "view_content\<^sub>m merge" where
+  "merge_view_content \<equiv> merge_prod merge_view_metadata merge_view_data"
+
+lift_definition hash_view :: "(view\<^sub>m, view\<^sub>h) hash" is
+  "hash_tree hash_view_content" .
+
+lift_definition blinding_of_view :: "view\<^sub>m blinding_of" is
+  "blinding_of_tree hash_view_content blinding_of_view_content" .
+
+lift_definition merge_view :: "view\<^sub>m merge" is
+  "merge_tree hash_view_content merge_view_content" .
+
+lemma merkle_view [locale_witness]: "merkle_interface hash_view blinding_of_view merge_view"
+  by transfer unfold_locales
+
+lemma hash_view_simps [simp]:
+  "hash_view (View\<^sub>m x) = 
+   View\<^sub>h (hash_blindable (hash_prod hash_view_content (hash_list hash_view)) x)"
+  by transfer(simp add: hash_rt_F\<^sub>m_def prod.map_comp hash_blindable_def blindable\<^sub>m.map_id)
+
+lemma blinding_of_view_iff [simp]:
+  "blinding_of_view (View\<^sub>m x) (View\<^sub>m y) \<longleftrightarrow>
+   blinding_of_blindable (hash_prod hash_view_content (hash_list hash_view))
+     (blinding_of_prod blinding_of_view_content (blinding_of_list blinding_of_view)) x y"
+  by transfer simp
+
+lemma blinding_of_view_induct [consumes 1, induct pred: blinding_of_view]:
+  assumes "blinding_of_view x y"
+    and "\<And>x y. blinding_of_blindable (hash_prod hash_view_content (hash_list hash_view))
+             (blinding_of_prod blinding_of_view_content (blinding_of_list (\<lambda>x y. blinding_of_view x y \<and> P x y))) x y
+         \<Longrightarrow> P (View\<^sub>m x) (View\<^sub>m y)"
+  shows "P x y"
+  using assms by transfer(rule blinding_of_tree.induct)
+
+lemma merge_view_simps [simp]:
+  "merge_view (View\<^sub>m x) (View\<^sub>m y) =
+   map_option View\<^sub>m (merge_rt_F\<^sub>m hash_view_content merge_view_content hash_view merge_view x y)"
+  by transfer simp
+
+end
+
+subsection \<open>Transaction trees as authenticated data structures\<close>
+
+type_synonym common_metadata\<^sub>h = "common_metadata blindable\<^sub>h"
+type_synonym common_metadata\<^sub>m = "(common_metadata, common_metadata) blindable\<^sub>m"
+
+type_synonym participant_metadata\<^sub>h = "participant_metadata blindable\<^sub>h"
+type_synonym participant_metadata\<^sub>m = "(participant_metadata, participant_metadata) blindable\<^sub>m"
+
+datatype transaction\<^sub>h = Transaction\<^sub>h 
+  (the_Transaction\<^sub>h: "((common_metadata\<^sub>h \<times>\<^sub>h participant_metadata\<^sub>h) \<times>\<^sub>h view\<^sub>h list\<^sub>h) blindable\<^sub>h")
+
+datatype transaction\<^sub>m = Transaction\<^sub>m 
+  (the_Transaction\<^sub>m: "((common_metadata\<^sub>m \<times>\<^sub>m participant_metadata\<^sub>m) \<times>\<^sub>m view\<^sub>m list\<^sub>m,
+    (common_metadata\<^sub>h \<times>\<^sub>h participant_metadata\<^sub>h) \<times>\<^sub>h view\<^sub>h list\<^sub>h) blindable\<^sub>m")
+
+abbreviation (input) hash_common_metadata :: "(common_metadata\<^sub>m, common_metadata\<^sub>h) hash" where
+  "hash_common_metadata \<equiv> hash_blindable id"
+abbreviation (input) blinding_of_common_metadata :: "common_metadata\<^sub>m blinding_of" where
+  "blinding_of_common_metadata \<equiv> blinding_of_blindable id (=)"
+abbreviation (input) merge_common_metadata :: "common_metadata\<^sub>m merge" where
+  "merge_common_metadata \<equiv> merge_blindable id merge_discrete"
+
+abbreviation (input) hash_participant_metadata :: "(participant_metadata\<^sub>m, participant_metadata\<^sub>h) hash" where
+  "hash_participant_metadata \<equiv> hash_blindable id"
+abbreviation (input) blinding_of_participant_metadata :: "participant_metadata\<^sub>m blinding_of" where
+  "blinding_of_participant_metadata \<equiv> blinding_of_blindable id (=)"
+abbreviation (input) merge_participant_metadata :: "participant_metadata\<^sub>m merge" where
+  "merge_participant_metadata \<equiv> merge_blindable id merge_discrete"
+
+locale transaction_merkle begin
+
+lemma iso_transaction\<^sub>h: "type_definition the_Transaction\<^sub>h Transaction\<^sub>h UNIV"
+  by unfold_locales simp_all
+
+setup_lifting iso_transaction\<^sub>h
+
+lemma Transaction\<^sub>h_transfer [transfer_rule]: includes lifting_syntax shows
+  "((=) ===> pcr_transaction\<^sub>h) id Transaction\<^sub>h"
+  by(simp add: transaction\<^sub>h.pcr_cr_eq cr_transaction\<^sub>h_def rel_fun_def)
+
+lemma iso_transaction\<^sub>m: "type_definition the_Transaction\<^sub>m Transaction\<^sub>m UNIV"
+  by unfold_locales simp_all
+
+setup_lifting iso_transaction\<^sub>m
+
+lemma Transaction\<^sub>m_transfer [transfer_rule]: includes lifting_syntax shows
+  "((=) ===> pcr_transaction\<^sub>m) id Transaction\<^sub>m"
+  by(simp add: transaction\<^sub>m.pcr_cr_eq cr_transaction\<^sub>m_def rel_fun_def)
+
+end
+
+code_datatype Transaction\<^sub>h
+code_datatype Transaction\<^sub>m
+
+context begin
+interpretation transaction_merkle .
+
+lift_definition hash_transaction :: "(transaction\<^sub>m, transaction\<^sub>h) hash" is
+  "hash_blindable (hash_prod (hash_prod hash_common_metadata hash_participant_metadata) (hash_list hash_view))" .
+
+lift_definition blinding_of_transaction :: "transaction\<^sub>m blinding_of" is
+  "blinding_of_blindable 
+     (hash_prod (hash_prod hash_common_metadata hash_participant_metadata) (hash_list hash_view))
+     (blinding_of_prod (blinding_of_prod blinding_of_common_metadata blinding_of_participant_metadata) (blinding_of_list blinding_of_view))" .
+
+lift_definition merge_transaction :: "transaction\<^sub>m merge" is
+  "merge_blindable
+     (hash_prod (hash_prod hash_common_metadata hash_participant_metadata) (hash_list hash_view))
+     (merge_prod (merge_prod merge_common_metadata merge_participant_metadata) (merge_list merge_view))" .
+
+lemma merkle_transaction [locale_witness]:
+  "merkle_interface hash_transaction blinding_of_transaction merge_transaction"
+  by transfer unfold_locales
+
+lemmas hash_transaction_simps [simp] = hash_transaction.abs_eq
+lemmas blinding_of_transaction_iff [simp] = blinding_of_transaction.abs_eq
+lemmas merge_transaction_simps [simp] = merge_transaction.abs_eq
+
+end
+
+interpretation transaction:
+  merkle_interface hash_transaction blinding_of_transaction merge_transaction 
+  by(rule merkle_transaction)
+
+subsection \<open>
+Constructing authenticated data structures for views
+\<close>
+
+context view_merkle begin
+
+type_synonym view' = "(view_metadata \<times> view_data) rose_tree"
+
+primrec from_view :: "view \<Rightarrow> view'" where
+  "from_view (View vm vd vs) = Tree ((vm, vd), map from_view vs)"
+
+primrec to_view :: "view' \<Rightarrow> view" where
+  "to_view (Tree x) = View (fst (fst x)) (snd (fst x)) (snd (map_prod id (map to_view) x))"
+
+lemma from_to_view [simp]: "from_view (to_view x) = x"
+  by(induction x)(clarsimp cong: map_cong)
+
+lemma to_from_view [simp]: "to_view (from_view x) = x"
+  by(induction x)(clarsimp cong: map_cong)
+
+lemma iso_view: "type_definition from_view to_view UNIV"
+  by unfold_locales simp_all
+
+setup_lifting iso_view
+
+definition View' :: "(view_metadata \<times> view_data) \<times> view list \<Rightarrow> view" where
+  "View' = (\<lambda>((vm, vd), vs). View vm vd vs)"
+
+lemma View_View': "View = (\<lambda>vm vd vs. View' ((vm, vd), vs))"
+  by(simp add: View'_def)
+
+lemma cr_view_Grp: "cr_view = Grp UNIV to_view"
+  by(simp add: cr_view_def Grp_def fun_eq_iff)(transfer, auto)
+
+lemma View'_transfer [transfer_rule]: includes lifting_syntax shows
+  "(rel_prod (=) (list_all2 pcr_view) ===> pcr_view) Tree View'"
+  by(simp add: view.pcr_cr_eq cr_view_Grp eq_alt prod.rel_Grp rose_tree.rel_Grp list.rel_Grp)
+    (auto simp add: Grp_def View'_def)
+
+end
+
+code_datatype View
+
+context begin
+interpretation view_merkle .
+
+abbreviation embed_view_content :: "view_metadata \<times> view_data \<Rightarrow> view_metadata\<^sub>m \<times> view_data\<^sub>m" where
+  "embed_view_content \<equiv> map_prod Unblinded Unblinded"
+
+lift_definition embed_view :: "view \<Rightarrow> view\<^sub>m" is "embed_source_tree embed_view_content" .
+
+lemma embed_view_simps [simp]:
+  "embed_view (View vm vd vs) = View\<^sub>m (Unblinded ((Unblinded vm, Unblinded vd), map embed_view vs))"
+  unfolding View_View' by transfer simp
+
+end
+
+context transaction_merkle begin
+
+primrec the_Transaction :: "transaction \<Rightarrow> (common_metadata \<times> participant_metadata) \<times> view list" where
+  "the_Transaction (Transaction cm pm views) = ((cm, pm), views)" for views
+
+definition Transaction' :: "(common_metadata \<times> participant_metadata) \<times> view list \<Rightarrow> transaction" where
+  "Transaction' = (\<lambda>((cm, pm), views). Transaction cm pm views)"
+
+lemma Transaction_Transaction': "Transaction = (\<lambda>cm pm views. Transaction' ((cm, pm), views))"
+  by(simp add: Transaction'_def)
+
+lemma the_Transaction_inverse [simp]: "Transaction' (the_Transaction x) = x" 
+  by(cases x)(simp add: Transaction'_def)
+
+lemma Transaction'_inverse [simp]: "the_Transaction (Transaction' x) = x"
+  by(simp add: Transaction'_def split_def)
+
+lemma iso_transaction: "type_definition the_Transaction Transaction' UNIV"
+  by unfold_locales simp_all
+
+setup_lifting iso_transaction
+
+lemma Transaction'_transfer [transfer_rule]: includes lifting_syntax shows
+  "((=) ===> pcr_transaction) id Transaction'"
+  by(simp add: transaction.pcr_cr_eq cr_transaction_def rel_fun_def)
+
+end
+
+code_datatype Transaction
+
+context begin
+interpretation transaction_merkle .
+
+lift_definition embed_transaction :: "transaction \<Rightarrow> transaction\<^sub>m" is
+  "Unblinded \<circ> map_prod (map_prod Unblinded Unblinded) (map embed_view)" .
+
+lemma embed_transaction_simps [simp]:
+  "embed_transaction (Transaction cm pm views) =
+   Transaction\<^sub>m (Unblinded ((Unblinded cm, Unblinded pm), map embed_view views))" 
+  for views unfolding Transaction_Transaction' by transfer simp
+
+end
+
+subsubsection \<open>Inclusion proof for the mediator\<close>
+
+primrec mediator_view :: "view \<Rightarrow> view\<^sub>m" where
+  "mediator_view (View vm vd vs) =
+   View\<^sub>m (Unblinded ((Unblinded vm, Blinded (Content vd)), map mediator_view vs))"
+
+primrec mediator_transaction_tree :: "transaction \<Rightarrow> transaction\<^sub>m" where
+  "mediator_transaction_tree (Transaction cm pm views) =
+   Transaction\<^sub>m (Unblinded ((Unblinded cm, Blinded (Content pm)), map mediator_view views))"
+  for views
+
+lemma blinding_of_mediator_view [simp]: "blinding_of_view (mediator_view view) (embed_view view)"
+  by(induction view)(auto simp add: list.rel_map intro!: list.rel_refl_strong)
+
+lemma blinding_of_mediator_transaction_tree:
+  "blinding_of_transaction (mediator_transaction_tree tt) (embed_transaction tt)"
+  by(cases tt)(auto simp add: list.rel_map intro: list.rel_refl_strong)
+
+subsubsection \<open>Inclusion proofs for participants\<close>
+
+text \<open>Next, we define a function for producing all transaction views from a given view,
+  and prove its properties.\<close>
+
+type_synonym view_path_elem = "(view_metadata \<times> view_data) blindable \<times> view list \<times> view list"
+type_synonym view_path = "view_path_elem list"
+type_synonym view_zipper = "view_path \<times> view"
+
+type_synonym view_path_elem\<^sub>m = "(view_metadata\<^sub>m \<times>\<^sub>m view_data\<^sub>m) \<times> view\<^sub>m list\<^sub>m \<times> view\<^sub>m list\<^sub>m"
+type_synonym view_path\<^sub>m = "view_path_elem\<^sub>m list"
+type_synonym view_zipper\<^sub>m = "view_path\<^sub>m \<times> view\<^sub>m"
+
+context begin
+interpretation view_merkle .
+
+lift_definition zipper_of_view :: "view \<Rightarrow> view_zipper" is zipper_of_tree .
+lift_definition view_of_zipper :: "view_zipper \<Rightarrow> view" is tree_of_zipper .
+
+lift_definition zipper_of_view\<^sub>m :: "view\<^sub>m \<Rightarrow> view_zipper\<^sub>m" is zipper_of_tree\<^sub>m .
+lift_definition view_of_zipper\<^sub>m :: "view_zipper\<^sub>m \<Rightarrow> view\<^sub>m" is tree_of_zipper\<^sub>m .
+
+lemma view_of_zipper\<^sub>m_Nil [simp]: "view_of_zipper\<^sub>m ([], t) = t"
+  by transfer simp
+
+lift_definition blind_view_path_elem :: "view_path_elem \<Rightarrow> view_path_elem\<^sub>m" is
+  "blind_path_elem embed_view_content hash_view_content" .
+
+lift_definition blind_view_path :: "view_path \<Rightarrow> view_path\<^sub>m" is
+  "blind_path embed_view_content hash_view_content" .
+
+lift_definition embed_view_path_elem :: "view_path_elem \<Rightarrow> view_path_elem\<^sub>m" is
+  "embed_path_elem embed_view_content" .
+
+lift_definition embed_view_path :: "view_path \<Rightarrow> view_path\<^sub>m" is
+  "embed_path embed_view_content" .
+
+lift_definition hash_view_path_elem :: "view_path_elem\<^sub>m \<Rightarrow> (view_content\<^sub>h \<times> view\<^sub>h list \<times> view\<^sub>h list)" is
+  "hash_path_elem hash_view_content" .
+
+lift_definition zippers_view :: "view_zipper \<Rightarrow> view_zipper\<^sub>m list" is
+  "zippers_rose_tree embed_view_content hash_view_content" .
+
+lemma embed_view_path_Nil [simp]: "embed_view_path [] = []"
+  by transfer(simp add: embed_path_def)
+
+lemma zippers_view_same_hash:
+  assumes "z \<in> set (zippers_view (p, t))"
+  shows "hash_view (view_of_zipper\<^sub>m z) = hash_view (view_of_zipper\<^sub>m (embed_view_path p, embed_view t))"
+  using assms by transfer(rule zippers_rose_tree_same_hash')
+
+lemma zippers_view_blinding_of:
+  assumes "z \<in> set (zippers_view (p, t))"
+  shows "blinding_of_view (view_of_zipper\<^sub>m z) (view_of_zipper\<^sub>m (blind_view_path p, embed_view t))"
+  using assms by transfer(rule zippers_rose_tree_blinding_of, unfold_locales)
+
+end
+
+primrec blind_view :: "view \<Rightarrow> view\<^sub>m" where
+  "blind_view (View vm vd subviews) = 
+   View\<^sub>m (Blinded (Content ((Content vm, Content vd), map (hash_view \<circ> embed_view) subviews)))"
+  for subviews
+
+lemma hash_blind_view: "hash_view (blind_view view) = hash_view (embed_view view)"
+  by(cases view) simp
+
+primrec blind_transaction :: "transaction \<Rightarrow> transaction\<^sub>m" where
+  "blind_transaction (Transaction cm pm views) = 
+   Transaction\<^sub>m (Blinded (Content ((Content cm, Content pm), map (hash_view \<circ> blind_view) views)))"
+  for views
+
+lemma hash_blind_transaction:
+  "hash_transaction (blind_transaction transaction) = hash_transaction (embed_transaction transaction)"
+  by(cases transaction)(simp add: hash_blind_view)
+
+
+typedecl participant
+consts recipients :: "view_metadata \<Rightarrow> participant list"
+
+fun view_recipients :: "view\<^sub>m \<Rightarrow> participant set" where
+  "view_recipients (View\<^sub>m (Unblinded ((Unblinded vm, vd), subviews))) = set (recipients vm)" for subviews
+| "view_recipients _ = {}" \<comment> \<open>Sane default case\<close>
+
+context fixes participant :: participant begin
+
+definition view_trees_for :: "view \<Rightarrow> view\<^sub>m list" where
+  "view_trees_for view =
+   map view_of_zipper\<^sub>m
+     (filter (\<lambda>(_, t). participant \<in> view_recipients t) 
+       (zippers_view ([], view)))"
+
+primrec transaction_views_for :: "transaction \<Rightarrow> transaction\<^sub>m list" where
+  "transaction_views_for (Transaction cm pm views) =
+   map (\<lambda>view\<^sub>m. Transaction\<^sub>m (Unblinded ((Unblinded cm, Unblinded pm), view\<^sub>m)))
+  (concat (map (\<lambda>(l, v, r). map (\<lambda>v\<^sub>m. map blind_view l @ [v\<^sub>m] @ map blind_view r) (view_trees_for v)) (splits views)))"
+  for views
+   
+lemma view_trees_for_same_hash:
+  "vt \<in> set (view_trees_for view) \<Longrightarrow> hash_view vt = hash_view (embed_view view)"
+  by(auto simp add: view_trees_for_def dest: zippers_view_same_hash)
+
+lemma transaction_views_for_same_hash:
+  "t\<^sub>m \<in> set (transaction_views_for t) \<Longrightarrow> hash_transaction t\<^sub>m = hash_transaction (embed_transaction t)"
+  by(cases t)(clarsimp simp add: splits_iff hash_blind_view view_trees_for_same_hash)
+
+definition transaction_projection_for :: "transaction \<Rightarrow> transaction\<^sub>m" where
+  "transaction_projection_for t =
+   (let tvs = transaction_views_for t
+    in if tvs = [] then blind_transaction t else the (transaction.Merge (set tvs)))"
+
+lemma transaction_projection_for_same_hash:
+  "hash_transaction (transaction_projection_for t) = hash_transaction (embed_transaction t)"
+proof(cases "transaction_views_for t = []")
+  case True thus ?thesis by(simp add: transaction_projection_for_def Let_def hash_blind_transaction)
+next
+  case False
+  then have "transaction.Merge (set (transaction_views_for t)) \<noteq> None"
+    by(intro transaction.Merge_defined)(auto simp add: transaction_views_for_same_hash)
+  with False show ?thesis
+    apply(clarsimp simp add: transaction_projection_for_def neq_Nil_conv simp del: transaction.Merge_insert)
+    apply(drule transaction.Merge_hash[symmetric], blast)
+    apply(auto intro: transaction_views_for_same_hash)
+    done
+qed
+
+lemma transaction_projection_for_upper:
+  assumes "t\<^sub>m \<in> set (transaction_views_for t)"
+  shows "blinding_of_transaction t\<^sub>m (transaction_projection_for t)"
+proof -
+  from assms have "transaction.Merge (set (transaction_views_for t)) \<noteq> None"
+    by(intro transaction.Merge_defined)(auto simp add: transaction_views_for_same_hash)
+  with assms show ?thesis
+    by(auto simp add: transaction_projection_for_def Let_def dest: transaction.Merge_upper)
+qed
+
+end
+
+end

formal/afp/ADS_Functor/Generic_ADS_Construction.thy

@@ -0,0 +1,469 @@
+(* Author: Andreas Lochbihler, Digital Asset
+   Author: Ognjen Maric, Digital Asset *)
+
+theory Generic_ADS_Construction imports
+  "Merkle_Interface"
+  "HOL-Library.BNF_Axiomatization"
+begin
+
+section \<open>Generic construction of authenticated data structures\<close>
+
+subsection \<open>Functors\<close>
+
+subsubsection \<open>Source functor\<close>
+
+text \<open>
+
+We want to allow ADSs of arbitrary ADTs, which we call "source trees". The ADTs we are interested in can
+in general be represented as the least fixpoints of some bounded natural (bi-)functor (BNF) \<open>('a, 'b) F\<close>, where
+@{typ 'a} is the type of "source" data, and @{typ 'b} is a recursion "handle".
+However, Isabelle's type system does not support higher kinds, necessary to parameterize our definitions
+over functors.
+Instead, we first develop a general theory of ADSs over an arbitrary, but fixed functor,
+and its least fixpoint. We show that the theory is compositional, in that the functor's least fixed point
+can then be reused as the "source" data of another functor.
+
+We start by defining the arbitrary fixed functor, its fixpoints, and showing how composition can be
+done. A higher-level explanation is found in the paper.
+\<close>
+
+
+bnf_axiomatization ('a, 'b) F [wits: "'a \<Rightarrow> ('a, 'b) F"]
+
+context notes [[typedef_overloaded]] begin
+datatype 'a T = T "('a, 'a T) F"
+end
+
+subsubsection \<open>Base Merkle functor\<close>
+
+text \<open>
+This type captures the ADS hashes.
+\<close>
+
+bnf_axiomatization ('a, 'b) F\<^sub>h [wits: "'a \<Rightarrow> ('a, 'b) F\<^sub>h"]
+
+text \<open>
+It intuitively contains mixed garbage and source values.
+The functor's recursive handle @{typ 'b} might contain partial garbage.
+\<close>
+
+
+text \<open>
+This type captures the ADS inclusion proofs.
+The functor \<open>('a, 'a', 'b, 'b') F\<^sub>m\<close> has all type variables doubled.
+This type represents all values including the information which parts are blinded.
+The original type variable @{typ 'a} now represents the source data, which for compositionality can contain blindable positions.
+The type @{typ 'b} is a recursive handle to inclusion sub-proofs (which can be partialy blinded). 
+The type @{typ 'a'} represent "hashes" of the source data in @{typ 'a}, i.e., a mix of source values and garbage.
+The type @{typ 'b'} is a recursive handle to ADS hashes of subtrees.
+
+The corresponding type of recursive authenticated trees is then a fixpoint of this functor.
+\<close>
+
+bnf_axiomatization ('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) F\<^sub>m [wits: "'a\<^sub>m \<Rightarrow> 'a\<^sub>h \<Rightarrow> 'b\<^sub>h \<Rightarrow> ('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) F\<^sub>m"]
+
+subsubsection \<open>Least fixpoint\<close>
+
+context notes [[typedef_overloaded]] begin
+datatype 'a\<^sub>h T\<^sub>h = T\<^sub>h "('a\<^sub>h, 'a\<^sub>h T\<^sub>h) F\<^sub>h"
+end
+
+context notes [[typedef_overloaded]] begin
+datatype ('a\<^sub>m, 'a\<^sub>h) T\<^sub>m = T\<^sub>m (the_T\<^sub>m: "('a\<^sub>m, 'a\<^sub>h, ('a\<^sub>m, 'a\<^sub>h) T\<^sub>m, 'a\<^sub>h T\<^sub>h) F\<^sub>m")
+end
+
+
+subsubsection \<open> Composition \<close>
+
+text \<open>
+Finally, we show how to compose two Merkle functors.
+For simplicity, we reuse @{typ \<open>('a, 'b) F\<close>} and @{typ \<open>'a T\<close>}.
+\<close>
+
+context notes [[typedef_overloaded]] begin
+
+datatype ('a, 'b) G = G "('a T, 'b) F"
+
+datatype ('a\<^sub>h, 'b\<^sub>h) G\<^sub>h = G\<^sub>h (the_G\<^sub>h: "('a\<^sub>h T\<^sub>h, 'b\<^sub>h) F\<^sub>h")
+
+datatype ('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) G\<^sub>m = G\<^sub>m (the_G\<^sub>m: "(('a\<^sub>m, 'a\<^sub>h) T\<^sub>m, 'a\<^sub>h T\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) F\<^sub>m")
+
+end
+
+
+subsection \<open>Root hash\<close>
+
+subsubsection \<open>Base functor\<close>
+
+text \<open>
+The root hash of an authenticated value is modelled as a blindable value of type @{typ "('a', 'b') F\<^sub>h"}.
+(Actually, we want to use an abstract datatype for root hashes, but we omit this distinction here for simplicity.)
+\<close>
+
+consts root_hash_F' :: "(('a\<^sub>h, 'a\<^sub>h, 'b\<^sub>h, 'b\<^sub>h) F\<^sub>m, ('a\<^sub>h, 'b\<^sub>h) F\<^sub>h) hash"
+  \<comment> \<open>Root hash operation where we assume that all atoms have already been replaced by root hashes.
+     This assumption is reflected in the equality of the type parameters of @{type F\<^sub>m} \<close>
+
+type_synonym ('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) hash_F =
+  "('a\<^sub>m, 'a\<^sub>h) hash \<Rightarrow> ('b\<^sub>m, 'b\<^sub>h) hash \<Rightarrow> (('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) F\<^sub>m, ('a\<^sub>h, 'b\<^sub>h) F\<^sub>h) hash"
+definition root_hash_F :: "('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) hash_F" where
+  "root_hash_F rha rhb = root_hash_F' \<circ> map_F\<^sub>m rha id rhb id"
+
+subsubsection \<open>Least fixpoint\<close>
+
+primrec root_hash_T' :: "(('a\<^sub>h, 'a\<^sub>h) T\<^sub>m, 'a\<^sub>h T\<^sub>h) hash" where
+  "root_hash_T' (T\<^sub>m x) = T\<^sub>h (root_hash_F' (map_F\<^sub>m id id root_hash_T' id x))"
+
+definition root_hash_T :: "('a\<^sub>m, 'a\<^sub>h) hash \<Rightarrow> (('a\<^sub>m, 'a\<^sub>h) T\<^sub>m, 'a\<^sub>h T\<^sub>h) hash" where
+  "root_hash_T rha = root_hash_T' \<circ> map_T\<^sub>m rha id"
+
+lemma root_hash_T_simps [simp]:
+  "root_hash_T rha (T\<^sub>m x) = T\<^sub>h (root_hash_F rha (root_hash_T rha) x)"
+  by(simp add: root_hash_T_def F\<^sub>m.map_comp root_hash_F_def T\<^sub>h.map_id0)
+
+subsubsection \<open>Composition\<close>
+
+primrec root_hash_G' :: "(('a\<^sub>h, 'a\<^sub>h, 'b\<^sub>h, 'b\<^sub>h) G\<^sub>m, ('a\<^sub>h, 'b\<^sub>h) G\<^sub>h) hash" where
+  "root_hash_G' (G\<^sub>m x) = G\<^sub>h (root_hash_F' (map_F\<^sub>m root_hash_T' id id id x))"
+
+definition root_hash_G :: "('a\<^sub>m, 'a\<^sub>h) hash \<Rightarrow> ('b\<^sub>m, 'b\<^sub>h) hash \<Rightarrow> (('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) G\<^sub>m, ('a\<^sub>h, 'b\<^sub>h) G\<^sub>h) hash" where
+  "root_hash_G rha rhb = root_hash_G' \<circ> map_G\<^sub>m rha id rhb id"
+
+lemma root_hash_G_unfold:
+  "root_hash_G rha rhb = G\<^sub>h \<circ> root_hash_F (root_hash_T rha) rhb \<circ> the_G\<^sub>m"
+  apply(rule ext)
+  subgoal for x
+    by(cases x)(simp add: root_hash_G_def fun_eq_iff root_hash_F_def root_hash_T_def F\<^sub>m.map_comp T\<^sub>m.map_comp o_def T\<^sub>h.map_id id_def[symmetric])
+  done
+
+lemma root_hash_G_simps [simp]:
+  "root_hash_G rha rhb (G\<^sub>m x) = G\<^sub>h (root_hash_F (root_hash_T rha) rhb x)"
+  by(simp add: root_hash_G_def root_hash_T_def F\<^sub>m.map_comp root_hash_F_def T\<^sub>h.map_id0)
+
+
+subsection \<open>Blinding relation\<close>
+
+text \<open>
+The blinding relation determines whether one ADS value is a blinding of another.
+\<close>
+
+subsubsection \<open> Blinding on the base functor (@{type F\<^sub>m}) \<close>
+
+type_synonym ('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) blinding_of_F =
+  "('a\<^sub>m, 'a\<^sub>h) hash \<Rightarrow> 'a\<^sub>m blinding_of \<Rightarrow> ('b\<^sub>m, 'b\<^sub>h) hash \<Rightarrow> 'b\<^sub>m blinding_of \<Rightarrow> ('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) F\<^sub>m blinding_of"
+
+\<comment> \<open> Computes whether a partially blinded ADS is a blinding of another one \<close>
+axiomatization blinding_of_F :: "('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) blinding_of_F" where
+  blinding_of_F_mono: "\<lbrakk> boa \<le> boa'; bob \<le> bob' \<rbrakk>
+    \<Longrightarrow> blinding_of_F rha boa rhb bob \<le> blinding_of_F rha boa' rhb bob'"
+    \<comment> \<open> Monotonicity must be unconditional (without the assumption @{text "blinding_of_on"}) 
+         such that we can justify the recursive definition for the least fixpoint. \<close>
+  and blinding_respects_hashes_F [locale_witness]:
+  "\<lbrakk> blinding_respects_hashes rha boa; blinding_respects_hashes rhb bob \<rbrakk>
+   \<Longrightarrow> blinding_respects_hashes (root_hash_F rha rhb) (blinding_of_F rha boa rhb bob)"
+  and blinding_of_on_F [locale_witness]:
+  "\<lbrakk> blinding_of_on A rha boa; blinding_of_on B rhb bob \<rbrakk>
+   \<Longrightarrow> blinding_of_on {x. set1_F\<^sub>m x \<subseteq> A \<and> set3_F\<^sub>m x \<subseteq> B} (root_hash_F rha rhb) (blinding_of_F rha boa rhb bob)"
+
+lemma blinding_of_F_mono_inductive:
+  assumes a: "\<And>x y. boa x y \<longrightarrow> boa' x y"
+    and b: "\<And>x y. bob x y \<longrightarrow> bob' x y"
+  shows "blinding_of_F rha boa rhb bob x y \<longrightarrow> blinding_of_F rha boa' rhb bob' x y"
+  using assms by(blast intro: blinding_of_F_mono[THEN predicate2D, OF predicate2I predicate2I])
+
+subsubsection \<open> Blinding on least fixpoints \<close>
+
+context
+  fixes rh :: "('a\<^sub>m, 'a\<^sub>h) hash"
+    and bo :: "'a\<^sub>m blinding_of"
+begin
+
+inductive blinding_of_T :: "('a\<^sub>m, 'a\<^sub>h) T\<^sub>m blinding_of" where
+  "blinding_of_T (T\<^sub>m x) (T\<^sub>m y)" if 
+  "blinding_of_F rh bo (root_hash_T rh) blinding_of_T x y"
+monos blinding_of_F_mono_inductive
+
+end
+
+lemma blinding_of_T_mono:
+  assumes "bo \<le> bo'"
+  shows "blinding_of_T rh bo \<le> blinding_of_T rh bo'"
+  by(rule predicate2I; erule blinding_of_T.induct)
+    (blast intro: blinding_of_T.intros blinding_of_F_mono[THEN predicate2D, OF assms, rotated -1])
+
+lemma blinding_of_T_root_hash:
+  assumes "bo \<le> vimage2p rh rh (=)"
+  shows "blinding_of_T rh bo \<le> vimage2p (root_hash_T rh) (root_hash_T rh) (=)"
+  apply(rule predicate2I vimage2pI)+
+  apply(erule blinding_of_T.induct)
+  apply simp
+  apply(drule blinding_respects_hashes_F[unfolded blinding_respects_hashes_def, THEN predicate2D, rotated -1])
+    apply(rule assms)
+   apply(blast intro: vimage2pI)
+  apply(simp add: vimage2p_def)
+  done
+
+lemma blinding_respects_hashes_T [locale_witness]:
+  "blinding_respects_hashes rh bo \<Longrightarrow> blinding_respects_hashes (root_hash_T rh) (blinding_of_T rh bo)"
+  unfolding blinding_respects_hashes_def by(rule blinding_of_T_root_hash)
+
+lemma blinding_of_on_T [locale_witness]:
+  assumes "blinding_of_on A rh bo"
+  shows "blinding_of_on {x. set1_T\<^sub>m x \<subseteq> A} (root_hash_T rh) (blinding_of_T rh bo)"
+  (is "blinding_of_on ?A ?h ?bo")
+proof -
+  interpret a: blinding_of_on A rh bo by fact
+  show ?thesis
+  proof
+    have "?bo x x \<and> (?bo x y \<longrightarrow> ?bo y z \<longrightarrow> ?bo x z) \<and> (?bo x y \<longrightarrow> ?bo y x \<longrightarrow> x = y)" 
+      if "x \<in> ?A" for x y z using that
+    proof(induction x arbitrary: y z)
+      case (T\<^sub>m x)
+      interpret blinding_of_on 
+        "{a. set1_F\<^sub>m a \<subseteq> A \<and> set3_F\<^sub>m a \<subseteq> set3_F\<^sub>m x}" 
+        "root_hash_F rh ?h" 
+        "blinding_of_F rh bo ?h ?bo"
+        apply(rule blinding_of_on_F[OF assms])
+        apply unfold_locales
+        subgoal using T\<^sub>m.IH T\<^sub>m.prems by(force simp add: eq_onp_def)
+        subgoal for a b c using T\<^sub>m.IH[of a b c] T\<^sub>m.prems by auto
+        subgoal for a b using T\<^sub>m.IH[of a b] T\<^sub>m.prems by auto
+        done
+      show ?case using T\<^sub>m.prems
+        apply(intro conjI)
+        subgoal by(auto intro: blinding_of_T.intros refl)
+        subgoal by(auto elim!: blinding_of_T.cases trans intro!: blinding_of_T.intros)
+        subgoal by(auto elim!: blinding_of_T.cases dest: antisym)
+        done
+    qed
+    then show "x \<in> ?A \<Longrightarrow> ?bo x x" 
+      and "\<lbrakk> ?bo x y; ?bo y z; x \<in> ?A \<rbrakk> \<Longrightarrow> ?bo x z"
+      and "\<lbrakk> ?bo x y; ?bo y x; x \<in> ?A \<rbrakk> \<Longrightarrow> x = y"
+      for x y z by blast+
+  qed
+qed
+
+lemmas blinding_of_T [locale_witness] = blinding_of_on_T[where A=UNIV, simplified]
+
+subsubsection \<open> Blinding on composition \<close>
+
+context
+  fixes rha :: "('a\<^sub>m, 'a\<^sub>h) hash"
+    and boa :: "'a\<^sub>m blinding_of"
+    and rhb :: "('b\<^sub>m, 'b\<^sub>h) hash"
+    and bob :: "'b\<^sub>m blinding_of"
+begin
+
+inductive blinding_of_G :: "('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) G\<^sub>m blinding_of" where
+  "blinding_of_G (G\<^sub>m x) (G\<^sub>m y)" if 
+  "blinding_of_F (root_hash_T rha) (blinding_of_T rha boa) rhb bob x y"
+
+lemma blinding_of_G_unfold:
+  "blinding_of_G = vimage2p the_G\<^sub>m the_G\<^sub>m (blinding_of_F (root_hash_T rha) (blinding_of_T rha boa) rhb bob)"
+  apply(rule ext)+
+  subgoal for x y by(cases x; cases y)(simp_all add: blinding_of_G.simps fun_eq_iff vimage2p_def)
+  done
+
+end
+
+lemma blinding_of_G_mono:
+  assumes "boa \<le> boa'" "bob \<le> bob'"
+  shows "blinding_of_G rha boa rhb bob \<le> blinding_of_G rha boa' rhb bob'"
+  unfolding blinding_of_G_unfold
+  by(rule vimage2p_mono' blinding_of_F_mono blinding_of_T_mono assms)+
+
+lemma blinding_of_G_root_hash:
+  assumes "boa \<le> vimage2p rha rha (=)" and "bob \<le> vimage2p rhb rhb (=)"
+  shows "blinding_of_G rha boa rhb bob \<le> vimage2p (root_hash_G rha rhb) (root_hash_G rha rhb) (=)"
+  unfolding blinding_of_G_unfold root_hash_G_unfold vimage2p_comp o_apply
+  apply(rule vimage2p_mono')
+  apply(rule order_trans)
+   apply(rule blinding_respects_hashes_F[unfolded blinding_respects_hashes_def])
+    apply(rule blinding_of_T_root_hash)
+    apply(rule assms)+
+  apply(rule vimage2p_mono')
+  apply(simp add: vimage2p_def)
+  done
+
+lemma blinding_of_on_G [locale_witness]:
+  assumes "blinding_of_on A rha boa" "blinding_of_on B rhb bob"
+  shows "blinding_of_on {x. set1_G\<^sub>m x \<subseteq> A \<and> set3_G\<^sub>m x \<subseteq> B} (root_hash_G rha rhb) (blinding_of_G rha boa rhb bob)"
+  (is "blinding_of_on ?A ?h ?bo")
+proof -
+  interpret a: blinding_of_on A rha boa by fact
+  interpret b: blinding_of_on B rhb bob by fact
+  interpret FT: blinding_of_on 
+    "{x. set1_F\<^sub>m x \<subseteq> {x. set1_T\<^sub>m x \<subseteq> A} \<and> set3_F\<^sub>m x \<subseteq> B}" 
+    "root_hash_F (root_hash_T rha) rhb"
+    "blinding_of_F (root_hash_T rha) (blinding_of_T rha boa) rhb bob"
+    ..
+  show ?thesis
+  proof
+    show "?bo \<le> vimage2p ?h ?h (=)"
+      using a.hash b.hash
+      by(rule blinding_of_G_root_hash)
+    show "?bo x x" if "x \<in> ?A" for x using that
+      by(cases x; hypsubst)(rule blinding_of_G.intros; rule FT.refl; auto)
+    show "?bo x z" if "?bo x y" "?bo y z" "x \<in> ?A" for x y z using that
+      by(fastforce elim!: blinding_of_G.cases intro!: blinding_of_G.intros elim!: FT.trans)
+    show "x = y" if "?bo x y" "?bo y x" "x \<in> ?A" for x y using that
+      by(clarsimp elim!: blinding_of_G.cases)(erule (1) FT.antisym; auto)
+  qed
+qed
+
+lemmas blinding_of_G [locale_witness] = blinding_of_on_G[where A=UNIV and B=UNIV, simplified]
+
+subsection \<open>Merging\<close>
+
+text \<open>Two Merkle values with the same root hash can be merged into a less blinded Merkle value.
+The operation is unspecified for trees with different root hashes.
+\<close>
+
+subsubsection \<open> Merging on the base functor \<close>
+
+axiomatization merge_F :: "('a\<^sub>m, 'a\<^sub>h) hash \<Rightarrow> 'a\<^sub>m merge \<Rightarrow> ('b\<^sub>m, 'b\<^sub>h) hash \<Rightarrow> 'b\<^sub>m merge 
+  \<Rightarrow> ('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) F\<^sub>m merge" where
+  merge_F_cong [fundef_cong]: 
+  "\<lbrakk> \<And>a b. a \<in> set1_F\<^sub>m x \<Longrightarrow> ma a b = ma' a b; \<And>a b. a \<in> set3_F\<^sub>m x \<Longrightarrow> mb a b = mb' a b \<rbrakk>
+   \<Longrightarrow> merge_F rha ma rhb mb x y = merge_F rha ma' rhb mb' x y"
+  and
+  merge_on_F [locale_witness]:
+  "\<lbrakk> merge_on A rha boa ma; merge_on B rhb bob mb \<rbrakk>
+  \<Longrightarrow> merge_on {x. set1_F\<^sub>m x \<subseteq> A \<and> set3_F\<^sub>m x \<subseteq> B} (root_hash_F rha rhb) (blinding_of_F rha boa rhb bob) (merge_F rha ma rhb mb)"
+
+lemmas merge_F [locale_witness] = merge_on_F[where A=UNIV and B=UNIV, simplified]
+
+subsubsection \<open> Merging on the least fixpoint \<close>
+
+lemma wfP_subterm_T: "wfP (\<lambda>x y. x \<in> set3_F\<^sub>m (the_T\<^sub>m y))"
+  apply(rule wfPUNIVI)
+  subgoal premises IH[rule_format] for P x
+    by(induct x)(auto intro: IH)
+  done
+
+lemma irrefl_subterm_T: "x \<in> set3_F\<^sub>m y \<Longrightarrow> y \<noteq> the_T\<^sub>m x"
+  using wfP_subterm_T by (auto simp: wfP_def elim!: wf_irrefl)
+
+context
+  fixes rh :: "('a\<^sub>m, 'a\<^sub>h) hash"
+  fixes m :: "'a\<^sub>m merge"
+begin
+
+function merge_T :: "('a\<^sub>m, 'a\<^sub>h) T\<^sub>m merge" where
+  "merge_T (T\<^sub>m x) (T\<^sub>m y) = map_option T\<^sub>m (merge_F rh m (root_hash_T rh) merge_T x y)"
+  by pat_completeness auto
+termination
+  apply(relation "{(x, y). x \<in> set3_F\<^sub>m (the_T\<^sub>m y)} <*lex*> {(x, y). x \<in> set3_F\<^sub>m (the_T\<^sub>m y)}")
+   apply(auto simp add: wfP_def[symmetric] wfP_subterm_T)
+  done
+
+lemma merge_on_T [locale_witness]:
+  assumes "merge_on A rh bo m"
+  shows "merge_on {x. set1_T\<^sub>m x \<subseteq> A} (root_hash_T rh) (blinding_of_T rh bo) merge_T"
+  (is "merge_on ?A ?h ?bo ?m")
+proof -
+  interpret a: merge_on A rh bo m by fact
+  show ?thesis
+  proof
+    have "(?h a = ?h b \<longrightarrow> (\<exists>ab. ?m a b = Some ab \<and> ?bo a ab \<and> ?bo b ab \<and> (\<forall>u. ?bo a u \<longrightarrow> ?bo b u \<longrightarrow> ?bo ab u))) \<and>
+      (?h a \<noteq> ?h b \<longrightarrow> ?m a b = None)"
+      if "a \<in> ?A" for a b using that unfolding mem_Collect_eq
+    proof(induction a arbitrary: b)
+      case (T\<^sub>m x y)
+      interpret merge_on "{y. set1_F\<^sub>m y \<subseteq> A \<and> set3_F\<^sub>m y \<subseteq> set3_F\<^sub>m x}"
+        "root_hash_F rh ?h" "blinding_of_F rh bo ?h ?bo" "merge_F rh m ?h ?m"
+      proof
+        fix a
+        assume a: "a \<in> set3_F\<^sub>m x"
+        with T\<^sub>m.prems have a': "set1_T\<^sub>m a \<subseteq> A" by auto
+
+        fix b
+        from T\<^sub>m.IH[OF a a', rule_format, of b]
+        show "root_hash_T rh a = root_hash_T rh b
+           \<Longrightarrow> \<exists>ab. merge_T a b = Some ab \<and> blinding_of_T rh bo a ab \<and> blinding_of_T rh bo b ab \<and>
+                    (\<forall>u. blinding_of_T rh bo a u \<longrightarrow> blinding_of_T rh bo b u \<longrightarrow> blinding_of_T rh bo ab u)"
+          and "root_hash_T rh a \<noteq> root_hash_T rh b \<Longrightarrow> merge_T a b = None"
+          by(auto dest: sym)
+      qed
+      show ?case using T\<^sub>m.prems
+        apply(intro conjI strip)
+        subgoal by(cases y)(auto dest!: join simp add: blinding_of_T.simps)
+        subgoal by(cases y)(auto dest!: undefined)
+        done
+    qed
+    then show
+      "?h a = ?h b \<Longrightarrow> \<exists>ab. ?m a b = Some ab \<and> ?bo a ab \<and> ?bo b ab \<and> (\<forall>u. ?bo a u \<longrightarrow> ?bo b u \<longrightarrow> ?bo ab u)"
+      "?h a \<noteq> ?h b \<Longrightarrow> ?m a b = None"
+      if "a \<in> ?A" for a b using that by blast+
+  qed
+qed
+
+lemmas merge_T [locale_witness] = merge_on_T[where A=UNIV, simplified]
+
+end
+
+lemma merge_T_cong [fundef_cong]:
+  assumes "\<And>a b. a \<in> set1_T\<^sub>m x \<Longrightarrow> m a b = m' a b"
+  shows "merge_T rh m x y = merge_T rh m' x y"
+  using assms
+  apply(induction x y rule: merge_T.induct)
+  apply simp
+  apply(rule arg_cong[where f="map_option _"])
+  apply(blast intro: merge_F_cong)
+  done
+
+subsubsection \<open> Merging and composition \<close>
+
+context
+  fixes rha :: "('a\<^sub>m, 'a\<^sub>h) hash"
+  fixes ma :: "'a\<^sub>m merge"
+  fixes rhb :: "('b\<^sub>m, 'b\<^sub>h) hash"
+  fixes mb :: "'b\<^sub>m merge"
+begin
+
+primrec merge_G :: "('a\<^sub>m, 'a\<^sub>h, 'b\<^sub>m, 'b\<^sub>h) G\<^sub>m merge" where
+  "merge_G (G\<^sub>m x) y' = (case y' of G\<^sub>m y \<Rightarrow>
+    map_option G\<^sub>m (merge_F (root_hash_T rha) (merge_T rha ma) rhb mb x y))"
+
+lemma merge_G_simps [simp]:
+  "merge_G (G\<^sub>m x) (G\<^sub>m y) = map_option G\<^sub>m (merge_F (root_hash_T rha) (merge_T rha ma) rhb mb x y)"
+  by(simp)
+
+declare merge_G.simps [simp del]
+
+lemma merge_on_G:
+  assumes a: "merge_on A rha boa ma" and b: "merge_on B rhb bob mb"
+  shows "merge_on {x. set1_G\<^sub>m x \<subseteq> A \<and> set3_G\<^sub>m x \<subseteq> B} (root_hash_G rha rhb) (blinding_of_G rha boa rhb bob) merge_G"
+  (is "merge_on ?A ?h ?bo ?m")
+proof -
+  interpret a: merge_on A rha boa ma by fact
+  interpret b: merge_on B rhb bob mb by fact
+  interpret F: merge_on 
+    "{x. set1_F\<^sub>m x \<subseteq> {x. set1_T\<^sub>m x \<subseteq> A} \<and> set3_F\<^sub>m x \<subseteq> B}"
+    "root_hash_F (root_hash_T rha) rhb"
+    "blinding_of_F (root_hash_T rha) (blinding_of_T rha boa) rhb bob"
+    "merge_F (root_hash_T rha) (merge_T rha ma) rhb mb"
+    ..
+  show ?thesis
+  proof
+    show "\<exists>ab. ?m a b = Some ab \<and> ?bo a ab \<and> ?bo b ab \<and> (\<forall>u. ?bo a u \<longrightarrow> ?bo b u \<longrightarrow> ?bo ab u)"
+      if "?h a = ?h b" "a \<in> ?A" for a b using that
+      by(cases a; cases b)(auto dest!: F.join simp add: blinding_of_G.simps)
+    show "?m a b = None" if "?h a \<noteq> ?h b" "a \<in> ?A" for a b using that
+      by(cases a; cases b)(auto dest!: F.undefined)
+  qed
+qed
+
+lemmas merge_G [locale_witness] = merge_on_G[where A=UNIV and B=UNIV, simplified]
+
+end
+
+lemma merge_G_cong [fundef_cong]: 
+  "\<lbrakk> \<And>a b. a \<in> set1_G\<^sub>m x \<Longrightarrow> ma a b = ma' a b; \<And>a b. a \<in> set3_G\<^sub>m x \<Longrightarrow> mb a b = mb' a b \<rbrakk>
+   \<Longrightarrow> merge_G rha ma rhb mb x y = merge_G rha ma' rhb mb' x y"
+  apply(cases x; cases y; simp)
+  apply(rule arg_cong[where f="map_option _"])
+  apply(blast intro: merge_F_cong merge_T_cong)
+  done
+
+end

formal/afp/ADS_Functor/Inclusion_Proof_Construction.thy

@@ -0,0 +1,430 @@
+(* Author: Andreas Lochbihler, Digital Asset
+   Author: Ognjen Maric, Digital Asset *)
+
+theory Inclusion_Proof_Construction imports
+  ADS_Construction
+begin
+
+primrec blind_blindable :: "('a\<^sub>m \<Rightarrow> 'a\<^sub>h) \<Rightarrow> ('a\<^sub>m, 'a\<^sub>h) blindable\<^sub>m \<Rightarrow> ('a\<^sub>m, 'a\<^sub>h) blindable\<^sub>m" where
+  "blind_blindable h (Blinded x) = Blinded x"
+| "blind_blindable h (Unblinded x) = Blinded (Content (h x))"
+
+lemma hash_blind_blindable [simp]: "hash_blindable h (blind_blindable h x) = hash_blindable h x"
+  by(cases x) simp_all
+
+subsection \<open>Inclusion proof construction for rose trees\<close>
+
+(************************************************************)
+subsubsection \<open> Hashing, embedding and blinding source trees \<close>
+(************************************************************)
+
+context fixes h :: "'a \<Rightarrow> 'a\<^sub>h" begin
+fun hash_source_tree :: "'a rose_tree \<Rightarrow> 'a\<^sub>h rose_tree\<^sub>h" where
+  "hash_source_tree (Tree (data, subtrees)) = Tree\<^sub>h (Content (h data, map hash_source_tree subtrees))"
+end
+
+context fixes e :: "'a \<Rightarrow> 'a\<^sub>m" begin
+fun embed_source_tree :: "'a rose_tree \<Rightarrow> ('a\<^sub>m, 'a\<^sub>h) rose_tree\<^sub>m" where
+  "embed_source_tree (Tree (data, subtrees)) =
+    Tree\<^sub>m (Unblinded (e data, map embed_source_tree subtrees))"
+end
+
+context fixes h :: "'a \<Rightarrow> 'a\<^sub>h" begin
+fun blind_source_tree :: "'a rose_tree \<Rightarrow> ('a\<^sub>m, 'a\<^sub>h) rose_tree\<^sub>m" where
+  "blind_source_tree (Tree (data, subtrees)) = Tree\<^sub>m (Blinded (Content (h data, map (hash_source_tree h) subtrees)))"
+end
+
+case_of_simps blind_source_tree_cases: blind_source_tree.simps
+
+fun is_blinded :: "('a\<^sub>m, 'a\<^sub>h) rose_tree\<^sub>m \<Rightarrow> bool" where
+  "is_blinded (Tree\<^sub>m (Blinded _)) = True"
+| "is_blinded _ = False"
+
+lemma hash_blinded_simp: "hash_tree h' (blind_source_tree h st) = hash_source_tree h st"
+  by(cases st rule: blind_source_tree.cases)(simp_all add: hash_rt_F\<^sub>m_def)
+
+lemma hash_embedded_simp:
+  "hash_tree h (embed_source_tree e st) = hash_source_tree (h \<circ> e) st"
+  by(induction st rule: embed_source_tree.induct)(simp add: hash_rt_F\<^sub>m_def)
+
+lemma blinded_embedded_same_hash:
+  "hash_tree h'' (blind_source_tree (h o e) st) = hash_tree h (embed_source_tree e st)"
+  by(simp add: hash_blinded_simp hash_embedded_simp)
+
+lemma blinding_blinds [simp]:
+  "is_blinded (blind_source_tree h t)"
+  by(simp add: blind_source_tree_cases split: rose_tree.split)
+
+lemma blinded_blinds_embedded: 
+  "blinding_of_tree h bo (blind_source_tree (h o e) st) (embed_source_tree e st)"
+  by(cases st rule: blind_source_tree.cases)(simp_all add: hash_embedded_simp)
+
+fun embed_hash_tree :: "'ha rose_tree\<^sub>h \<Rightarrow> ('a, 'ha) rose_tree\<^sub>m" where
+  "embed_hash_tree (Tree\<^sub>h h) = Tree\<^sub>m (Blinded h)"
+
+
+(************************************************************)
+subsubsection \<open>Auxiliary definitions: selectors and list splits\<close>
+(************************************************************)
+
+fun children :: "'a rose_tree \<Rightarrow> 'a rose_tree list" where
+  "children (Tree (data, subtrees)) = subtrees"
+
+fun children\<^sub>m :: "('a, 'a\<^sub>h) rose_tree\<^sub>m \<Rightarrow> ('a, 'a\<^sub>h) rose_tree\<^sub>m list" where
+  "children\<^sub>m (Tree\<^sub>m (Unblinded (data, subtrees))) = subtrees"
+| "children\<^sub>m _ = undefined"
+
+fun splits :: "'a list \<Rightarrow> ('a list \<times> 'a \<times> 'a list) list" where
+  "splits [] = []"
+| "splits (x#xs) = ([], x, xs) # map (\<lambda>(l, y, r). (x # l, y, r)) (splits xs)"
+
+lemma splits_iff: "(l, a, r) \<in> set (splits ll) = (ll = l @ a # r)"
+  by(induction ll arbitrary: l a r)(auto simp add: Cons_eq_append_conv)
+
+(************************************************************)
+subsubsection \<open> Zippers \<close>
+(************************************************************)
+
+text \<open> Zippers provide a neat representation of tree-like ADSs when they have only a single 
+  unblinded subtree. The zipper path provides the "inclusion proof" that the unblinded subtree is
+  included in a larger structure. \<close>
+
+type_synonym 'a path_elem = "'a \<times> 'a rose_tree list \<times> 'a rose_tree list"
+type_synonym 'a path = "'a path_elem list"
+type_synonym 'a zipper = "'a path \<times> 'a rose_tree"
+
+definition zipper_of_tree :: "'a rose_tree \<Rightarrow> 'a zipper" where
+  "zipper_of_tree t \<equiv> ([], t)"
+               
+fun tree_of_zipper :: "'a zipper \<Rightarrow> 'a rose_tree" where
+  "tree_of_zipper ([], t) = t"
+| "tree_of_zipper ((a, l, r) # z, t) = tree_of_zipper (z, (Tree (a, (l @ t # r))))"
+
+case_of_simps tree_of_zipper_cases: tree_of_zipper.simps
+
+lemma tree_of_zipper_id[iff]: "tree_of_zipper (zipper_of_tree t) = t"
+  by(simp add: zipper_of_tree_def)
+
+fun zipper_children :: "'a zipper \<Rightarrow> 'a zipper list" where
+  "zipper_children (p, Tree (a, ts)) = map (\<lambda>(l, t, r). ((a, l, r) # p, t)) (splits ts)"
+
+lemma zipper_children_same_tree:
+  assumes "z' \<in> set (zipper_children z)"
+  shows "tree_of_zipper z' = tree_of_zipper z"
+proof-
+  obtain p a ts where z: "z = (p, Tree (a, ts))"
+    using assms
+    by(cases z rule: zipper_children.cases) (simp_all)
+
+  then obtain l t r where ltr: "z' = ((a, l, r) # p, t)" and "(l, t, r) \<in> set (splits ts)"
+    using assms
+    by(auto)
+
+  with z show ?thesis
+    by(simp add: splits_iff)
+qed
+
+type_synonym ('a\<^sub>m, 'a\<^sub>h) path_elem\<^sub>m = "'a\<^sub>m \<times> ('a\<^sub>m, 'a\<^sub>h) rose_tree\<^sub>m list \<times> ('a\<^sub>m, 'a\<^sub>h) rose_tree\<^sub>m list"
+type_synonym ('a\<^sub>m, 'a\<^sub>h) path\<^sub>m = "('a\<^sub>m, 'a\<^sub>h) path_elem\<^sub>m list"
+type_synonym ('a\<^sub>m, 'a\<^sub>h) zipper\<^sub>m = "('a\<^sub>m, 'a\<^sub>h) path\<^sub>m \<times> ('a\<^sub>m, 'a\<^sub>h) rose_tree\<^sub>m"
+
+definition zipper_of_tree\<^sub>m :: "('a\<^sub>m, 'a\<^sub>h) rose_tree\<^sub>m \<Rightarrow> ('a\<^sub>m, 'a\<^sub>h) zipper\<^sub>m" where
+  "zipper_of_tree\<^sub>m t \<equiv> ([], t)"
+
+fun tree_of_zipper\<^sub>m :: "('a\<^sub>m, 'a\<^sub>h) zipper\<^sub>m \<Rightarrow> ('a\<^sub>m, 'a\<^sub>h) rose_tree\<^sub>m" where
+  "tree_of_zipper\<^sub>m ([], t) = t"
+| "tree_of_zipper\<^sub>m ((m, l, r) # z, t) = tree_of_zipper\<^sub>m (z, Tree\<^sub>m (Unblinded (m, l @ t # r)))"
+
+lemma tree_of_zipper\<^sub>m_append:
+  "tree_of_zipper\<^sub>m (p @ p', t) = tree_of_zipper\<^sub>m (p', tree_of_zipper\<^sub>m (p, t))"
+  by(induction p arbitrary: p' t) auto
+
+fun zipper_children\<^sub>m :: "('a\<^sub>m, 'a\<^sub>h) zipper\<^sub>m \<Rightarrow> ('a\<^sub>m, 'a\<^sub>h) zipper\<^sub>m list" where
+  "zipper_children\<^sub>m (p, Tree\<^sub>m (Unblinded (a, ts))) = map (\<lambda>(l, t, r). ((a, l, r) # p, t)) (splits ts) "
+| "zipper_children\<^sub>m _ = []"
+
+lemma zipper_children_same_tree\<^sub>m:
+  assumes "z' \<in> set (zipper_children\<^sub>m z)"
+  shows "tree_of_zipper\<^sub>m z' = tree_of_zipper\<^sub>m z"
+proof-
+  obtain p a ts where z: "z = (p, Tree\<^sub>m (Unblinded (a, ts)))"
+    using assms
+    by(cases z rule: zipper_children\<^sub>m.cases) (simp_all)
+
+  then obtain l t r where ltr: "z' = ((a, l, r) # p, t)" and "(l, t, r) \<in> set (splits ts)"
+    using assms
+    by(auto)
+
+  with z show ?thesis
+    by(simp add: splits_iff)
+qed
+
+fun blind_path_elem :: "('a \<Rightarrow> 'a\<^sub>m) \<Rightarrow> ('a\<^sub>m \<Rightarrow> 'a\<^sub>h) \<Rightarrow> 'a path_elem \<Rightarrow> ('a\<^sub>m, 'a\<^sub>h) path_elem\<^sub>m"  where
+  "blind_path_elem e h (x, l, r) = (e x, map (blind_source_tree (h \<circ> e)) l, map (blind_source_tree (h \<circ> e)) r)"
+
+case_of_simps blind_path_elem_cases: blind_path_elem.simps
+
+definition blind_path :: "('a \<Rightarrow> 'a\<^sub>m) \<Rightarrow> ('a\<^sub>m \<Rightarrow> 'a\<^sub>h) \<Rightarrow> 'a path \<Rightarrow> ('a\<^sub>m, 'a\<^sub>h) path\<^sub>m" where
+  "blind_path e h \<equiv> map (blind_path_elem e h)"
+
+fun embed_path_elem :: "('a \<Rightarrow> 'a\<^sub>m) \<Rightarrow> 'a path_elem \<Rightarrow> ('a\<^sub>m, 'a\<^sub>h) path_elem\<^sub>m" where
+  "embed_path_elem e (d, l, r) = (e d, map (embed_source_tree e) l, map (embed_source_tree e) r)"
+
+definition embed_path :: "('a \<Rightarrow> 'a\<^sub>m) \<Rightarrow> 'a path \<Rightarrow> ('a\<^sub>m, 'a\<^sub>h) path\<^sub>m" where
+  "embed_path embed_elem \<equiv> map (embed_path_elem embed_elem)"
+
+lemma hash_tree_of_zipper_same_path: 
+  "hash_tree h (tree_of_zipper\<^sub>m (p, v)) = hash_tree h (tree_of_zipper\<^sub>m (p, v'))
+        \<longleftrightarrow> hash_tree h v = hash_tree h v'"
+  by(induction p arbitrary: v v')(auto simp add: hash_rt_F\<^sub>m_def)
+
+fun hash_path_elem :: "('a\<^sub>m \<Rightarrow> 'a\<^sub>h) \<Rightarrow> ('a\<^sub>m, 'a\<^sub>h) path_elem\<^sub>m \<Rightarrow> ('a\<^sub>h \<times> 'a\<^sub>h rose_tree\<^sub>h list \<times> 'a\<^sub>h rose_tree\<^sub>h list)" where
+  "hash_path_elem h (e, l, r) = (h e, map (hash_tree h) l, map (hash_tree h) r)"
+
+lemma hash_view_zipper_eqI:
+  "\<lbrakk> hash_list (hash_path_elem h) p = hash_list (hash_path_elem h') p';
+    hash_tree h v = hash_tree h' v' \<rbrakk> \<Longrightarrow>
+    hash_tree h (tree_of_zipper\<^sub>m (p, v)) = hash_tree h' (tree_of_zipper\<^sub>m (p', v'))"
+  by(induction p arbitrary: p' v v')(auto simp add: hash_rt_F\<^sub>m_def)
+
+lemma blind_embed_path_same_hash:
+  "hash_tree h (tree_of_zipper\<^sub>m (blind_path e h p, t)) = hash_tree h (tree_of_zipper\<^sub>m (embed_path e p, t))"
+proof -
+  have "hash_path_elem h \<circ> blind_path_elem e h = hash_path_elem h \<circ> embed_path_elem e"
+    by(clarsimp simp add: hash_blinded_simp hash_embedded_simp fun_eq_iff intro!: arg_cong2[where f=hash_source_tree, OF _ refl])
+  then show ?thesis
+    by(intro hash_view_zipper_eqI)(simp_all add: embed_path_def blind_path_def list.map_comp)
+qed
+
+lemma tree_of_embed_commute:
+  "tree_of_zipper\<^sub>m (embed_path e p, embed_source_tree e t) = embed_source_tree e (tree_of_zipper (p, t))"
+  by(induction "(p, t)" arbitrary: p t rule: tree_of_zipper.induct)(simp_all add: embed_path_def)
+
+lemma childz_same_tree:
+  "(l, t, r) \<in> set (splits ts) \<Longrightarrow>
+  tree_of_zipper\<^sub>m (embed_path e p, embed_source_tree e (Tree (d, ts))) 
+  = tree_of_zipper\<^sub>m (embed_path e ((d, l, r) # p), embed_source_tree e t)"
+  by(simp add: tree_of_embed_commute splits_iff del: embed_source_tree.simps)
+
+lemma blinding_of_same_path:
+  assumes bo: "blinding_of_on UNIV h bo"
+  shows
+  "blinding_of_tree h bo (tree_of_zipper\<^sub>m (p, t)) (tree_of_zipper\<^sub>m (p, t'))
+  \<longleftrightarrow> blinding_of_tree h bo t t'"
+proof -
+  interpret a: blinding_of_on UNIV h bo by fact
+  interpret tree: blinding_of_on UNIV "hash_tree h" "blinding_of_tree h bo" ..
+  show ?thesis
+    by(induction p arbitrary: t t')(auto simp add: list_all2_append list.rel_refl a.refl tree.refl)
+qed
+
+lemma zipper_children_size_change [termination_simp]: "(a, b) \<in> set (zipper_children (p, v)) \<Longrightarrow> size b < size v"
+  by(cases v)(clarsimp simp add: splits_iff Set.image_iff)
+
+
+subsection \<open>All zippers of a rose tree\<close>
+
+context fixes e :: "'a \<Rightarrow> 'a\<^sub>m" and h :: "'a\<^sub>m \<Rightarrow> 'a\<^sub>h" begin
+
+fun zippers_rose_tree :: "'a zipper \<Rightarrow> ('a\<^sub>m, 'a\<^sub>h) zipper\<^sub>m list" where
+  "zippers_rose_tree (p, t) = (blind_path e h p, embed_source_tree e t) # 
+    concat (map zippers_rose_tree (zipper_children (p, t)))"
+
+end
+
+lemmas [simp del] = zippers_rose_tree.simps zipper_children.simps
+
+lemma zippers_rose_tree_same_hash':
+  assumes "z \<in> set (zippers_rose_tree e h (p, t))"
+  shows "hash_tree h (tree_of_zipper\<^sub>m z) = 
+    hash_tree h (tree_of_zipper\<^sub>m (embed_path e p, embed_source_tree e t))"
+  using assms(1)
+proof(induction "(p, t)" arbitrary: p t rule: zippers_rose_tree.induct)
+  case (1 p t)
+  from "1.prems"[unfolded zippers_rose_tree.simps] 
+  consider (find) "z = (blind_path e h p, embed_source_tree e t)"
+    | (rec) x ts l t' r where "t = Tree (x, ts)" "(l, t', r) \<in> set (splits ts)" "z \<in> set (zippers_rose_tree e h ((x, l, r) # p, t'))"
+    by(cases t)(auto simp add: zipper_children.simps) 
+  then show ?case
+  proof cases
+    case rec
+    then show ?thesis
+      apply(subst "1.hyps"[of "(x, l, r) # p" "t'"])
+      apply(simp_all add: rev_image_eqI zipper_children.simps)
+      by (metis (no_types) childz_same_tree comp_apply embed_source_tree.simps rec(2))
+  qed(simp add: blind_embed_path_same_hash)
+qed
+
+lemma zippers_rose_tree_blinding_of:
+  assumes "blinding_of_on UNIV h bo" 
+    and z: "z \<in> set (zippers_rose_tree e h (p, t))"
+  shows "blinding_of_tree h bo (tree_of_zipper\<^sub>m z) (tree_of_zipper\<^sub>m (blind_path e h p, embed_source_tree e t))"
+  using z
+proof(induction "(p, t)" arbitrary: p t rule: zippers_rose_tree.induct)
+  case (1 p t)
+
+  interpret a: blinding_of_on UNIV h bo by fact
+  interpret rt: blinding_of_on UNIV "hash_tree h" "blinding_of_tree h bo" ..
+
+  from "1.prems"[unfolded zippers_rose_tree.simps] 
+  consider (find) "z = (blind_path e h p, embed_source_tree e t)"
+    | (rec) x ts l t' r where "t = Tree (x, ts)" "(l, t', r) \<in> set (splits ts)" "z \<in> set (zippers_rose_tree e h ((x, l, r) # p, t'))"
+    by(cases t)(auto simp add: zipper_children.simps) 
+  then show ?case 
+  proof cases
+    case find
+    then show ?thesis by(simp add: rt.refl)
+  next
+    case rec
+    then have "blinding_of_tree h bo 
+      (tree_of_zipper\<^sub>m z) 
+      (tree_of_zipper\<^sub>m (blind_path e h ((x, l, r) # p), embed_source_tree e t'))"
+      by(intro 1)(simp add: rev_image_eqI zipper_children.simps)
+    also have "blinding_of_tree h bo 
+      (tree_of_zipper\<^sub>m (blind_path e h ((x, l, r) # p), embed_source_tree e t'))
+      (tree_of_zipper\<^sub>m (blind_path e h p, embed_source_tree e (Tree (x, ts))))"
+      using rec
+      by(simp add: blind_path_def splits_iff blinding_of_same_path[OF assms(1)] a.refl list_all2_append list_all2_same list.rel_map blinded_blinds_embedded rt.refl)
+    finally (rt.trans) show ?thesis using rec by simp
+  qed
+qed
+
+lemma zippers_rose_tree_neq_Nil: "zippers_rose_tree e h (p, t) \<noteq> []"
+  by(simp add: zippers_rose_tree.simps)
+
+lemma (in comp_fun_idem) fold_set_union:
+  assumes "finite A" "finite B"
+  shows "Finite_Set.fold f z (A \<union> B) = Finite_Set.fold f (Finite_Set.fold f z A) B"
+  using assms(2,1) by induct simp_all
+
+context merkle_interface begin
+
+lemma comp_fun_idem_merge: "comp_fun_idem (\<lambda>x yo. yo \<bind> m x)"
+  apply(unfold_locales; clarsimp simp add: fun_eq_iff split: bind_split)
+  subgoal by (metis assoc bind.bind_lunit bind.bind_lzero idem option.distinct(1))
+  subgoal by (simp add: join)
+  done
+
+interpretation merge: comp_fun_idem "\<lambda>x yo. yo \<bind> m x" by(rule comp_fun_idem_merge)
+
+definition Merge :: "'a\<^sub>m set \<Rightarrow> 'a\<^sub>m option" where
+  "Merge A = (if A = {} \<or> infinite A then None else Finite_Set.fold (\<lambda>x yo. yo \<bind> m x) (Some (SOME x. x \<in> A)) A)"
+
+lemma Merge_empty [simp]: "Merge {} = None"
+  by(simp add: Merge_def)
+
+lemma Merge_infinite [simp]: "infinite A \<Longrightarrow> Merge A = None"
+  by(simp add: Merge_def)
+
+lemma Merge_cong_start:
+  "Finite_Set.fold (\<lambda>x yo. yo \<bind> m x) (Some x) A = Finite_Set.fold (\<lambda>x yo. yo \<bind> m x) (Some y) A" (is "?lhs = ?rhs")
+  if "x \<in> A" "y \<in> A" "finite A"
+proof -
+  have "?lhs = Finite_Set.fold (\<lambda>x yo. yo \<bind> m x) (Some x) (insert y A)" using that by(simp add: insert_absorb)
+  also have "\<dots> = Finite_Set.fold (\<lambda>x yo. yo \<bind> m x) (m x y) A" using that 
+    by(simp only: merge.fold_insert_idem2)(simp add: commute)
+  also have "\<dots> = Finite_Set.fold (\<lambda>x yo. yo \<bind> m x) (Some y) (insert x A)" using that
+    by(simp only: merge.fold_insert_idem2)(simp)
+  also have "\<dots> = ?rhs" using that by(simp add: insert_absorb)
+  finally show ?thesis .
+qed
+
+lemma Merge_insert [simp]: "Merge (insert x A) = (if A = {} then Some x else Merge A \<bind> m x)" (is "?lhs = ?rhs")
+proof(cases "finite A \<and> A \<noteq> {}")
+  case True
+  then have "?lhs = Finite_Set.fold (\<lambda>x yo. yo \<bind> m x) (Some (SOME x. x \<in> A)) (insert x A)" 
+    unfolding Merge_def by(subst Merge_cong_start[where y="SOME x. x \<in> A", OF someI])(auto intro: someI)
+  also have "\<dots> = ?rhs" using True by(simp add: Merge_def)
+  finally show ?thesis .
+qed(auto simp add: Merge_def idem)
+
+lemma Merge_insert_alt:
+  "Merge (insert x A) = Finite_Set.fold (\<lambda>x yo. yo \<bind> m x) (Some x) A" (is "?lhs = ?rhs") if "finite A"
+proof -
+  have "?lhs = Finite_Set.fold (\<lambda>x yo. yo \<bind> m x) (Some x) (insert x A)" using that
+    unfolding Merge_def by(subst Merge_cong_start[where y=x, OF someI]) auto
+  also have "\<dots> = ?rhs" using that by(simp only: merge.fold_insert_idem2)(simp add: idem)
+  finally show ?thesis .
+qed
+
+lemma Merge_None [simp]: "Finite_Set.fold (\<lambda>x yo. yo \<bind> m x) None A = None"
+proof(cases "finite A")
+  case True
+  then show ?thesis by(induction) auto
+qed simp
+
+lemma Merge_union: 
+  "Merge (A \<union> B) = (if A = {} then Merge B else if B = {} then Merge A else (Merge A \<bind> (\<lambda>a. Merge B \<bind> m a)))"
+  (is "?lhs = ?rhs")
+proof(cases "finite (A \<union> B) \<and> A \<noteq> {} \<and> B \<noteq> {}")
+  case True
+  then have "?lhs = Finite_Set.fold (\<lambda>x yo. yo \<bind> m x) (Some (SOME x. x \<in> B)) (B \<union> A)"
+    unfolding Merge_def by(subst Merge_cong_start[where y="SOME x. x \<in> B", OF someI])(auto intro: someI simp add: Un_commute)
+  also have "\<dots> = Finite_Set.fold (\<lambda>x yo. yo \<bind> m x) (Merge B) A" using True 
+    by(simp add: Merge_def merge.fold_set_union)
+  also have "\<dots> = Merge A \<bind> (\<lambda>a. Merge B \<bind> m a)"
+  proof(cases "Merge B")
+    case (Some b)
+    thus ?thesis using True 
+      by simp(subst Merge_insert_alt[symmetric]; simp add: commute; metis commute)
+  qed simp
+  finally show ?thesis using True by simp
+qed auto
+
+lemma Merge_upper:
+  assumes m: "Merge A = Some x" and y: "y \<in> A"
+  shows "bo y x"
+proof -
+  have "Merge A = Merge (insert y A)" using y by(simp add: insert_absorb)
+  also have "\<dots> = Merge A \<bind> m y" using y by auto
+  finally have "m y x = Some x" using m by simp
+  thus ?thesis by(simp add: bo_def)
+qed
+
+lemma Merge_least:
+  assumes m: "Merge A = Some x" and u[rule_format]: "\<forall>a\<in>A. bo a u"
+  shows "bo x u"
+proof -
+  define a where "a \<equiv> SOME x. x \<in> A"
+  from m have A: "finite A" "A \<noteq> {}" 
+    and *: "Finite_Set.fold (\<lambda>x yo. yo \<bind> m x) (Some a) A = Some x" 
+    by(auto simp add: Merge_def a_def split: if_splits)
+  from A have "bo a u" by(auto intro: someI u simp add: a_def)
+  with A * u show ?thesis
+  proof(induction A arbitrary: a)
+    case (insert x A)
+    then show ?case
+      by(cases "m x a"; cases "A = {}"; simp only: merge.fold_insert_idem2; simp)(auto simp add: join)
+  qed simp
+qed
+
+lemma Merge_defined:
+  assumes "finite A" "A \<noteq> {}" "\<forall>a\<in>A. \<forall>b \<in> A. h a = h b"
+  shows "Merge A \<noteq> None"
+proof
+  define a where "a \<equiv> SOME a. a \<in> A"
+  have a: "a \<in> A" unfolding a_def using assms by(auto intro: someI)
+  hence ha: "\<forall>b \<in> A. h b = h a" using assms by blast
+
+  assume m: "Merge A = None"
+  hence "Finite_Set.fold (\<lambda>x yo. yo \<bind> m x) (Some a) A = None"
+    using assms by(simp add: Merge_def a_def)
+  with assms(1) show False using ha
+  proof(induction arbitrary: a)
+    case (insert x A)
+    thus ?case
+      apply(cases "m x a"; use nothing in \<open>simp only: merge.fold_insert_idem2\<close>)
+       apply(simp add: merge_respects_hashes)
+      apply(fastforce simp add: join vimage2p_def dest: hash[THEN predicate2D])
+      done
+  qed simp
+qed
+
+lemma Merge_hash:
+  assumes "Merge A = Some x" "a \<in> A"
+  shows "h a = h x"
+  using Merge_upper[OF assms] hash by(auto simp add: vimage2p_def)
+
+end
+
+end

formal/afp/ADS_Functor/Merkle_Interface.thy

@@ -0,0 +1,299 @@
+(* Author: Andreas Lochbihler, Digital Asset
+   Author: Ognjen Maric, Digital Asset *)
+
+theory Merkle_Interface
+imports
+  Main
+  "HOL-Library.Conditional_Parametricity"
+  "HOL-Library.Monad_Syntax"
+begin
+
+alias vimage2p = BNF_Def.vimage2p
+alias Grp = BNF_Def.Grp
+alias setl = Basic_BNFs.setl
+alias setr = Basic_BNFs.setr
+alias fsts = Basic_BNFs.fsts
+alias snds = Basic_BNFs.snds
+
+attribute_setup locale_witness = \<open>Scan.succeed Locale.witness_add\<close>
+
+lemma vimage2p_mono': "R \<le> S \<Longrightarrow> vimage2p f g R \<le> vimage2p f g S"
+  by(auto simp add: vimage2p_def le_fun_def)
+
+lemma vimage2p_map_rel_prod: 
+  "vimage2p (map_prod f g) (map_prod f' g') (rel_prod A B) = rel_prod (vimage2p f f' A) (vimage2p g g' B)"
+  by(simp add: vimage2p_def prod.rel_map)
+
+lemma vimage2p_map_list_all2:
+  "vimage2p (map f) (map g) (list_all2 A) = list_all2 (vimage2p f g A)"
+  by(simp add: vimage2p_def list.rel_map)
+
+lemma equivclp_least:
+  assumes le: "r \<le> s" and s: "equivp s"
+  shows "equivclp r \<le> s"
+  apply(rule predicate2I)
+  subgoal by(induction rule: equivclp_induct)(auto 4 3 intro: equivp_reflp[OF s] equivp_transp[OF s] equivp_symp[OF s] le[THEN predicate2D])
+  done
+
+lemma reflp_eq_onp: "reflp R \<longleftrightarrow> eq_onp (\<lambda>x. True) \<le> R"
+  by(auto simp add: reflp_def eq_onp_def)
+
+lemma eq_onpE:
+  assumes "eq_onp P x y"
+  obtains "x = y" "P y"
+  using assms by(auto simp add: eq_onp_def)
+
+lemma case_unit_parametric [transfer_rule]: "rel_fun A (rel_fun (=) A) case_unit case_unit"
+  by(simp add: rel_fun_def split: unit.split)
+
+
+(************************************************************)
+section \<open>Authenticated Data Structures\<close>
+(************************************************************)
+
+(************************************************************)
+subsection \<open>Interface\<close>
+(************************************************************)
+
+(************************************************************)
+subsubsection \<open> Types \<close>
+(************************************************************)
+
+type_synonym ('a\<^sub>m, 'a\<^sub>h) hash = "'a\<^sub>m \<Rightarrow> 'a\<^sub>h" \<comment> \<open>Type of hash operation\<close>
+type_synonym 'a\<^sub>m blinding_of = "'a\<^sub>m \<Rightarrow> 'a\<^sub>m \<Rightarrow> bool"
+type_synonym 'a\<^sub>m merge = "'a\<^sub>m \<Rightarrow> 'a\<^sub>m \<Rightarrow> 'a\<^sub>m option" \<comment> \<open> merging that can fail for values with different hashes\<close>
+
+(************************************************************)
+subsubsection \<open> Properties \<close>
+(************************************************************)
+
+locale merkle_interface =
+  fixes h :: "('a\<^sub>m, 'a\<^sub>h) hash"
+    and bo :: "'a\<^sub>m blinding_of"
+    and m :: "'a\<^sub>m merge"
+  assumes merge_respects_hashes: "h a = h b \<longleftrightarrow> (\<exists>ab. m a b = Some ab)"
+    and idem: "m a a = Some a"
+    and commute: "m a b = m b a"
+    and assoc: "m a b \<bind> m c = m b c \<bind> m a"
+    and bo_def: "bo a b \<longleftrightarrow> m a b = Some b"
+begin
+
+lemma reflp: "reflp bo"
+  unfolding bo_def by(rule reflpI)(simp add: idem)
+
+lemma antisymp: "antisymp bo"
+  unfolding bo_def by(rule antisympI)(simp add: commute)
+
+lemma transp: "transp bo"
+  apply(rule transpI)
+  subgoal for x y z using assoc[of x y z] by(simp add: commute bo_def)
+  done
+
+lemma hash: "bo \<le> vimage2p h h (=)"
+  unfolding bo_def by(auto simp add: vimage2p_def merge_respects_hashes)
+
+lemma join: "m a b = Some ab \<longleftrightarrow> bo a ab \<and> bo b ab \<and> (\<forall>u. bo a u \<longrightarrow> bo b u \<longrightarrow> bo ab u)"
+  unfolding bo_def
+  by (smt Option.bind_cong bind.bind_lunit commute idem merkle_interface.assoc merkle_interface_axioms)
+
+text \<open>The equivalence closure of the blinding relation are the equivalence classes of the hash function (the kernel).\<close>
+
+lemma equivclp_blinding_of: "equivclp bo = vimage2p h h (=)" (is "?lhs = ?rhs")
+proof(rule antisym)
+  show "?lhs \<le> ?rhs" by(rule equivclp_least[OF hash])(rule equivp_vimage2p[OF identity_equivp])
+  show "?rhs \<le> ?lhs" unfolding vimage2p_def
+  proof(rule predicate2I)
+    fix x y
+    assume "h x = h y"
+    then obtain xy where "m x y = Some xy" unfolding merge_respects_hashes ..
+    hence "bo x xy" "bo y xy" unfolding join by blast+
+    hence "equivclp bo x xy" "equivclp bo xy y" by(blast)+
+    thus "equivclp bo x y" by(rule equivclp_trans)
+  qed
+qed
+
+end
+
+(************************************************************)
+subsection \<open> Auxiliary definitions \<close>
+(************************************************************)
+
+text \<open> Directly proving that an interface satisfies the specification of a Merkle interface as given
+above is difficult. Instead, we provide several layers of auxiliary definitions that can easily be
+proved layer-by-layer. 
+
+In particular, proving that an interface on recursive datatypes is a Merkle interface requires
+induction. As the induction hypothesis only applies to a subset of values of a type, we add
+auxiliary definitions equipped with an explicit set @{term A} of values to which the definition
+applies. Once the induction proof is complete, we can typically instantiate @{term A} with @{term
+UNIV}. In particular, in the induction proof for a layer, we can assume that properties for the
+earlier layers hold for \<^emph>\<open>all\<close> values, not just those in the induction hypothesis.
+\<close>
+
+(************************************************************)
+subsubsection \<open> Blinding \<close>
+(************************************************************)
+locale blinding_respects_hashes =
+  fixes h :: "('a\<^sub>m, 'a\<^sub>h) hash"
+    and bo :: "'a\<^sub>m blinding_of"
+  assumes hash: "bo \<le> vimage2p h h (=)"
+begin
+
+lemma blinding_hash_eq: "bo x y \<Longrightarrow> h x = h y"
+  by(drule hash[THEN predicate2D])(simp add: vimage2p_def)
+
+end
+
+locale blinding_of_on =
+  blinding_respects_hashes h bo
+    for A :: "'a\<^sub>m set"
+    and h :: "('a\<^sub>m, 'a\<^sub>h) hash"
+    and bo :: "'a\<^sub>m blinding_of"
+  + assumes refl: "x \<in> A \<Longrightarrow> bo x x"
+    and trans: "\<lbrakk> bo x y; bo y z; x \<in> A \<rbrakk> \<Longrightarrow> bo x z"
+    and antisym: "\<lbrakk> bo x y; bo y x; x \<in> A \<rbrakk> \<Longrightarrow> x = y"
+begin
+
+lemma refl_pointfree: "eq_onp (\<lambda>x. x \<in> A) \<le> bo"
+  by(auto elim!: eq_onpE intro: refl)
+
+lemma blinding_respects_hashes: "blinding_respects_hashes h bo" ..
+lemmas hash = hash
+
+lemma trans_pointfree: "eq_onp (\<lambda>x. x \<in> A) OO bo OO bo \<le> bo"
+  by(auto elim!: eq_onpE intro: trans)
+
+lemma antisym_pointfree: "inf (eq_onp (\<lambda>x. x \<in> A) OO bo) bo\<inverse>\<inverse> \<le> (=)"
+  by(auto elim!: eq_onpE dest: antisym)
+
+end
+
+(************************************************************)
+subsubsection \<open> Merging \<close>
+(************************************************************)
+
+text \<open> In general, we prove the properties of blinding before the properties of merging. Thus,
+  in the following definitions we assume that the blinding properties already hold on @{term UNIV}.
+  The @{term Ball} restricts the argument of the merge operation on which induction will be done. \<close>
+
+locale merge_on =
+  blinding_of_on UNIV h bo
+    for A :: "'a\<^sub>m set"
+    and h :: "('a\<^sub>m, 'a\<^sub>h) hash"
+    and bo :: "'a\<^sub>m blinding_of" 
+    and m :: "'a\<^sub>m merge" +
+  assumes join: "\<lbrakk> h a = h b; a \<in> A \<rbrakk> 
+      \<Longrightarrow> \<exists>ab. m a b = Some ab \<and> bo a ab \<and> bo b ab \<and> (\<forall>u. bo a u \<longrightarrow> bo b u \<longrightarrow> bo ab u)"
+    and undefined: "\<lbrakk> h a \<noteq> h b; a \<in> A \<rbrakk> \<Longrightarrow> m a b = None"
+begin
+
+lemma same: "a \<in> A \<Longrightarrow> m a a = Some a"
+  using join[of a a] refl[of a] by(auto 4 3 intro: antisym)
+
+lemma blinding_of_antisym_on: "blinding_of_on UNIV h bo" ..
+
+lemma transp: "transp bo"
+  by(auto intro: transpI trans)
+
+lemmas hash = hash
+  and refl = refl
+  and antisym = antisym[OF _ _ UNIV_I]
+
+lemma respects_hashes:
+  "a \<in> A \<Longrightarrow> h a = h b \<longleftrightarrow> (\<exists>ab. m a b = Some ab)"
+  using join undefined
+  by fastforce
+
+lemma join':
+  "a \<in> A \<Longrightarrow> \<forall>ab. m a b = Some ab \<longleftrightarrow> bo a ab \<and> bo b ab \<and> (\<forall>u. bo a u \<longrightarrow> bo b u \<longrightarrow> bo ab u)"
+  using join undefined
+  by (metis (full_types) hash local.antisym option.distinct(1) option.sel predicate2D vimage2p_def)
+
+lemma merge_on_subset:
+  "B \<subseteq> A \<Longrightarrow> merge_on B h bo m"
+  by unfold_locales (auto dest: same join undefined)
+
+end
+
+subsection \<open> Interface equality \<close>
+
+text \<open> Here, we prove that the auxiliary definitions specify the same interface as the original ones.\<close>
+
+lemma merkle_interface_aux: "merkle_interface h bo m = merge_on UNIV h bo m" 
+  (is "?lhs = ?rhs")
+proof
+  show ?rhs if ?lhs
+  proof
+    interpret merkle_interface h bo m by(fact that)
+    show "bo \<le> vimage2p h h (=)" by(fact hash)
+    show "bo x x" for x using reflp by(simp add: reflp_def)
+    show "bo x z" if "bo x y" "bo y z" for x y z using transp that by(rule transpD)
+    show "x = y" if "bo x y" "bo y x" for x y using antisymp that by(rule antisympD)
+    show "\<exists>ab. m a b = Some ab \<and> bo a ab \<and> bo b ab \<and> (\<forall>u. bo a u \<longrightarrow> bo b u \<longrightarrow> bo ab u)" if "h a = h b" for a b
+      using that by(simp add: merge_respects_hashes join)
+    show "m a b = None" if "h a \<noteq> h b" for a b using that by(simp add: merge_respects_hashes)
+  qed
+
+  show ?lhs if ?rhs
+  proof
+    interpret merge_on UNIV h bo m by(fact that)
+    show eq: "h a = h b \<longleftrightarrow> (\<exists>ab. m a b = Some ab)" for a b by(simp add: respects_hashes)
+    show idem: "m a a = Some a" for a by(simp add: same)
+    show commute: "m a b = m b a" for a b 
+      using join[of a b] join[of b a] undefined antisym by(cases "m a b") force+
+    have undefined_partitioned: "m a c = None" if "m a b = None" "m b c = Some bc" for a b c bc
+      using that eq by (metis option.distinct(1) option.exhaust)
+    have merge_twice: "m a b = Some c \<Longrightarrow> m a c = Some c" for a b c by (simp add: join')
+    show "m a b \<bind> m c = m b c \<bind> m a" for a b c
+    proof(simp split: Option.bind_split; safe)
+      show "None = m a d" if "m a b = None" "m b c = Some d" for d using that
+        by(metis undefined_partitioned merge_twice)
+      show "m c d = None" if "m a b = Some d" "m b c = None" for d using that
+        by(metis commute merge_twice undefined_partitioned)
+    next
+      fix ab bc
+      assume assms: "m a b = Some ab" "m b c = Some bc"
+      then obtain cab and abc where cab: "m c ab = Some cab" and abc: "m a bc = Some abc"
+        using eq[THEN iffD2, OF exI] eq[THEN iffD1] by (metis merge_twice)
+      thus "m c ab = m a bc" using assms
+        by(clarsimp simp add: join')(metis UNIV_I abc cab local.antisym local.trans)
+    qed
+    show "bo a b \<longleftrightarrow> m a b = Some b" for a b using idem join' by auto
+  qed
+qed
+
+lemma merkle_interfaceI [locale_witness]:
+  assumes "merge_on UNIV h bo m"
+  shows "merkle_interface h bo m"
+  using assms unfolding merkle_interface_aux by auto
+
+lemma (in merkle_interface) merkle_interfaceD: "merge_on UNIV h bo m"
+  using merkle_interface_aux[of h bo m, symmetric]
+  by simp unfold_locales
+
+subsection \<open> Parametricity rules \<close>
+
+context includes lifting_syntax begin
+parametric_constant le_fun_parametric[transfer_rule]: le_fun_def
+parametric_constant vimage2p_parametric[transfer_rule]: vimage2p_def
+parametric_constant blinding_respects_hashes_parametric_aux: blinding_respects_hashes_def
+
+lemma blinding_respects_hashes_parametric [transfer_rule]:
+  "((A1 ===> A2) ===> (A1 ===> A1 ===> (\<longleftrightarrow>)) ===> (\<longleftrightarrow>))
+   blinding_respects_hashes blinding_respects_hashes"
+  if [transfer_rule]: "bi_unique A2" "bi_total A1"
+  by(rule blinding_respects_hashes_parametric_aux that le_fun_parametric | simp add: rel_fun_eq)+
+
+parametric_constant blinding_of_on_axioms_parametric [transfer_rule]:
+  blinding_of_on_axioms_def[folded Ball_def, unfolded le_fun_def le_bool_def eq_onp_def relcompp.simps, simplified]
+parametric_constant blinding_of_on_parametric [transfer_rule]: blinding_of_on_def
+parametric_constant antisymp_parametric[transfer_rule]: antisymp_def
+parametric_constant transp_parametric[transfer_rule]: transp_def
+
+parametric_constant merge_on_axioms_parametric [transfer_rule]: merge_on_axioms_def
+parametric_constant merge_on_parametric[transfer_rule]: merge_on_def
+
+parametric_constant merkle_interface_parametric[transfer_rule]: merkle_interface_def
+end
+
+end

formal/afp/ADS_Functor/document/root.tex

@@ -0,0 +1,78 @@
+\documentclass[11pt,a4paper]{article}
+\usepackage[T1]{fontenc}
+\usepackage{isabelle,isabellesym}
+
+% further packages required for unusual symbols (see also
+% isabellesym.sty), use only when needed
+
+%\usepackage{amssymb}
+  %for \<leadsto>, \<box>, \<diamond>, \<sqsupset>, \<mho>, \<Join>,
+  %\<lhd>, \<lesssim>, \<greatersim>, \<lessapprox>, \<greaterapprox>,
+  %\<triangleq>, \<yen>, \<lozenge>
+
+%\usepackage{eurosym}
+  %for \<euro>
+
+%\usepackage[only,bigsqcap]{stmaryrd}
+  %for \<Sqinter>
+
+%\usepackage{eufrak}
+  %for \<AA> ... \<ZZ>, \<aa> ... \<zz> (also included in amssymb)
+
+%\usepackage{textcomp}
+  %for \<onequarter>, \<onehalf>, \<threequarters>, \<degree>, \<cent>,
+  %\<currency>
+
+% this should be the last package used
+\usepackage{pdfsetup}
+
+% urls in roman style, theory text in math-similar italics
+\urlstyle{rm}
+\isabellestyle{it}
+
+% for uniform font size
+%\renewcommand{\isastyle}{\isastyleminor}
+
+
+\begin{document}
+
+\title{Authenticated Data Structures as Functors}
+\author{Andreas Lochbihler \qquad Ognjen Maric \\[1em] Digital Asset}
+
+\maketitle
+
+\begin{abstract}
+  Authenticated data structures allow several systems to convince each other that they are referring to the same data structure,
+  even if each of them knows only a part of the data structure.
+  Using inclusion proofs, knowledgable systems can selectively share their knowledge with other systems
+  and the latter can verify the authenticity of what is being shared.
+
+  In this paper, we show how to modularly define authenticated data structures, their inclusion proofs, and operations thereon
+  as datatypes in Isabelle/HOL, using a shallow embedding.
+  Modularity allows us to construct complicated trees from reusable building blocks, which we call Merkle functors.
+  Merkle functors include sums, products, and function spaces and are closed under composition and least fixpoints.
+
+  As a practical application, we model the hierarchical transactions of Canton,
+  a practical interoperability protocol for distributed ledgers, as authenticated data structures.
+  This is a first step towards formalizing the Canton protocol and verifying its integrity and security guarantees.
+\end{abstract}
+
+
+\tableofcontents
+
+% sane default for proof documents
+\parindent 0pt\parskip 0.5ex
+
+% generated text of all theories
+\input{session}
+
+% optional bibliography
+%\bibliographystyle{abbrv}
+%\bibliography{root}
+
+\end{document}
+
+%%% Local Variables:
+%%% mode: latex
+%%% TeX-master: t
+%%% End:

formal/afp/AI_Planning_Languages_Semantics/Error_Monad_Add.thy

@@ -0,0 +1,52 @@
+theory Error_Monad_Add
+imports
+  "Certification_Monads.Check_Monad"
+  "Show.Show_Instances"
+begin
+  (* TODO: Move *)  
+  abbreviation "assert_opt \<Phi> \<equiv> if \<Phi> then Some () else None"  
+
+  definition "lift_opt m e \<equiv> case m of Some x \<Rightarrow> Error_Monad.return x | None \<Rightarrow> Error_Monad.error e"
+    
+  lemma lift_opt_simps[simp]: 
+    "lift_opt None e = error e"
+    "lift_opt (Some v) e = return v"
+    by (auto simp: lift_opt_def)
+  
+  (* TODO: Move *)  
+  lemma reflcl_image_iff[simp]: "R\<^sup>=``S = S\<union>R``S" by blast 
+    
+  named_theorems return_iff
+      
+  lemma bind_return_iff[return_iff]: "Error_Monad.bind m f = Inr y \<longleftrightarrow> (\<exists>x. m = Inr x \<and> f x = Inr y)"
+    by auto
+    
+  lemma lift_opt_return_iff[return_iff]: "lift_opt m e = Inr x \<longleftrightarrow> m=Some x"
+    unfolding lift_opt_def by (auto split: option.split)
+      
+  lemma check_return_iff[return_iff]: "check \<Phi> e = Inr uu \<longleftrightarrow> \<Phi>"    
+    by (auto simp: check_def)
+  
+  
+  lemma check_simps[simp]:  
+    "check True e = succeed"
+    "check False e = error e"
+    by (auto simp: check_def)
+        
+  lemma Let_return_iff[return_iff]: "(let x=v in f x) = Inr w \<longleftrightarrow> f v = Inr w" by simp
+
+  
+  abbreviation ERR :: "shows \<Rightarrow> (unit \<Rightarrow> shows)" where "ERR s \<equiv> \<lambda>_. s"
+  abbreviation ERRS :: "String.literal \<Rightarrow> (unit \<Rightarrow> shows)" where "ERRS s \<equiv> ERR (shows s)"
+  
+  
+  lemma error_monad_bind_split: "P (bind m f) \<longleftrightarrow> (\<forall>v. m = Inl v \<longrightarrow> P (Inl v)) \<and> (\<forall>v. m = Inr v \<longrightarrow> P (f v))"
+    by (cases m) auto
+  
+  lemma error_monad_bind_split_asm: "P (bind m f) \<longleftrightarrow> \<not> (\<exists>x. m = Inl x \<and> \<not> P (Inl x) \<or> (\<exists>x. m = Inr x \<and> \<not> P (f x)))"
+    by (cases m) auto
+  
+  lemmas error_monad_bind_splits =error_monad_bind_split error_monad_bind_split_asm
+  
+  
+end

formal/afp/AI_Planning_Languages_Semantics/Lifschitz_Consistency.thy

@@ -0,0 +1,416 @@
+section \<open>Soundness theorem for the STRIPS semantics\<close>
+text \<open>We prove the soundness theorem according to ~\cite{lifschitz1987semantics}.\<close>
+
+theory Lifschitz_Consistency
+imports PDDL_STRIPS_Semantics
+begin
+
+
+text \<open>States are modeled as valuations of our underlying predicate logic.\<close>
+type_synonym state = "(predicate\<times>object list) valuation"
+
+context ast_domain begin
+  text \<open>An action is a partial function from states to states. \<close>
+  type_synonym action = "state \<rightharpoonup> state"
+
+  text \<open>The Isabelle/HOL formula @{prop \<open>f s = Some s'\<close>} means
+    that \<open>f\<close> is applicable in state \<open>s\<close>, and the result is \<open>s'\<close>. \<close>
+
+  text \<open>Definition B (i)--(iv) in Lifschitz's paper~\cite{lifschitz1987semantics}\<close>
+
+  fun is_NegPredAtom where
+    "is_NegPredAtom (Not x) = is_predAtom x" | "is_NegPredAtom _ = False"
+
+  definition "close_eq s = (\<lambda>predAtm p xs \<Rightarrow> s (p,xs) | Eq a b \<Rightarrow> a=b)"
+
+  lemma close_eq_predAtm[simp]: "close_eq s (predAtm p xs) \<longleftrightarrow> s (p,xs)"
+    by (auto simp: close_eq_def)                    
+
+  lemma close_eq_Eq[simp]: "close_eq s (Eq a b) \<longleftrightarrow> a=b"
+    by (auto simp: close_eq_def)
+
+
+  abbreviation entail_eq :: "state \<Rightarrow> object atom formula \<Rightarrow> bool" (infix "\<Turnstile>\<^sub>=" 55)
+    where "entail_eq s f \<equiv> close_eq s \<Turnstile> f"
+
+
+  fun sound_opr :: "ground_action \<Rightarrow> action \<Rightarrow> bool" where
+    "sound_opr (Ground_Action pre (Effect add del)) f \<longleftrightarrow>
+      (\<forall>s. s \<Turnstile>\<^sub>= pre \<longrightarrow>
+        (\<exists>s'. f s = Some s' \<and> (\<forall>atm. is_predAtom atm \<and> atm \<notin> set del \<and> s \<Turnstile>\<^sub>= atm \<longrightarrow> s' \<Turnstile>\<^sub>= atm)
+              \<and>  (\<forall>atm. is_predAtom atm \<and> atm \<notin> set add \<and> s \<Turnstile>\<^sub>= Not atm \<longrightarrow> s' \<Turnstile>\<^sub>= Not atm)
+              \<and> (\<forall>fmla. fmla \<in> set add \<longrightarrow> s' \<Turnstile>\<^sub>= fmla)
+              \<and> (\<forall>fmla. fmla \<in> set del \<and> fmla \<notin> set add \<longrightarrow> s' \<Turnstile>\<^sub>= (Not fmla))
+              ))
+        \<and> (\<forall>fmla\<in>set add. is_predAtom fmla)"
+
+  lemma sound_opr_alt:
+    "sound_opr opr f =
+      ((\<forall>s. s \<Turnstile>\<^sub>= (precondition opr) \<longrightarrow>
+          (\<exists>s'. f s = (Some s')
+                \<and> (\<forall>atm. is_predAtom atm \<and> atm \<notin> set(dels (effect opr)) \<and> s \<Turnstile>\<^sub>= atm \<longrightarrow> s' \<Turnstile>\<^sub>= atm)
+                \<and> (\<forall>atm. is_predAtom atm \<and> atm \<notin> set (adds (effect opr)) \<and> s \<Turnstile>\<^sub>= Not atm \<longrightarrow> s' \<Turnstile>\<^sub>= Not atm)
+                \<and> (\<forall>atm. atm \<in> set(adds (effect opr)) \<longrightarrow> s' \<Turnstile>\<^sub>= atm)
+                \<and> (\<forall>fmla. fmla \<in> set (dels (effect opr)) \<and> fmla \<notin> set(adds (effect opr)) \<longrightarrow> s' \<Turnstile>\<^sub>= (Not fmla))
+                \<and> (\<forall>a b. s \<Turnstile>\<^sub>= Atom (Eq a b) \<longrightarrow> s' \<Turnstile>\<^sub>= Atom (Eq a b))
+                \<and> (\<forall>a b. s \<Turnstile>\<^sub>= Not (Atom (Eq a b)) \<longrightarrow> s' \<Turnstile>\<^sub>= Not (Atom (Eq a b)))
+                ))
+        \<and> (\<forall>fmla\<in>set(adds (effect opr)). is_predAtom fmla))"
+    by (cases "(opr,f)" rule: sound_opr.cases) auto
+
+  text \<open>Definition B (v)--(vii) in  Lifschitz's paper~\cite{lifschitz1987semantics}\<close>
+definition sound_system
+    :: "ground_action set
+      \<Rightarrow> world_model
+      \<Rightarrow> state
+      \<Rightarrow> (ground_action \<Rightarrow> action)
+      \<Rightarrow> bool"
+    where
+    "sound_system \<Sigma> M\<^sub>0 s\<^sub>0 f \<longleftrightarrow>
+      ((\<forall>fmla\<in>close_world M\<^sub>0. s\<^sub>0  \<Turnstile>\<^sub>= fmla)
+      \<and> wm_basic M\<^sub>0
+      \<and> (\<forall>\<alpha>\<in>\<Sigma>. sound_opr \<alpha> (f \<alpha>)))"
+
+  text \<open>Composing two actions\<close>
+  definition compose_action :: "action \<Rightarrow> action \<Rightarrow> action" where
+    "compose_action f1 f2 x = (case f2 x of Some y \<Rightarrow> f1 y | None \<Rightarrow> None)"
+
+  text \<open>Composing a list of actions\<close>
+  definition compose_actions :: "action list \<Rightarrow> action" where
+    "compose_actions fs \<equiv> fold compose_action fs Some"
+
+  text \<open>Composing a list of actions satisfies some natural lemmas: \<close>
+  lemma compose_actions_Nil[simp]:
+    "compose_actions [] = Some" unfolding compose_actions_def by auto
+
+  lemma compose_actions_Cons[simp]:
+    "f s = Some s' \<Longrightarrow> compose_actions (f#fs) s = compose_actions fs s'"
+  proof -
+    interpret monoid_add compose_action Some
+      apply unfold_locales
+      unfolding compose_action_def
+      by (auto split: option.split)
+    assume "f s = Some s'"
+    then show ?thesis
+      unfolding compose_actions_def
+      by (simp add: compose_action_def fold_plus_sum_list_rev)
+  qed
+
+  text \<open>Soundness Theorem in Lifschitz's paper~\cite{lifschitz1987semantics}.\<close>
+theorem STRIPS_sema_sound:
+  assumes "sound_system \<Sigma> M\<^sub>0 s\<^sub>0 f"
+    \<comment> \<open>For a sound system \<open>\<Sigma>\<close>\<close>
+  assumes "set \<alpha>s \<subseteq> \<Sigma>"
+    \<comment> \<open>And a plan \<open>\<alpha>s\<close>\<close>
+  assumes "ground_action_path M\<^sub>0 \<alpha>s M'"
+    \<comment> \<open>Which is accepted by the system, yielding result \<open>M'\<close>
+          (called \<open>R(\<alpha>s)\<close> in Lifschitz's paper~\cite{lifschitz1987semantics}.)\<close>
+  obtains s'
+    \<comment> \<open>We have that \<open>f(\<alpha>s)\<close> is applicable
+          in initial state, yielding state \<open>s'\<close> (called \<open>f\<^sub>\<alpha>\<^sub>s(s\<^sub>0)\<close> in Lifschitz's paper~\cite{lifschitz1987semantics}.)\<close>
+  where "compose_actions (map f \<alpha>s) s\<^sub>0 = Some s'"
+    \<comment> \<open>The result world model \<open>M'\<close> is satisfied in state \<open>s'\<close>\<close>
+    and "\<forall>fmla\<in>close_world M'. s' \<Turnstile>\<^sub>= fmla"
+proof -
+  have "(valuation M' \<Turnstile> fmla)" if "wm_basic M'" "fmla\<in>M'" for fmla
+    using that apply (induction fmla)
+    by (auto simp: valuation_def wm_basic_def split: atom.split)
+  have "\<exists>s'. compose_actions (map f \<alpha>s) s\<^sub>0 = Some s' \<and> (\<forall>fmla\<in>close_world M'. s' \<Turnstile>\<^sub>= fmla)"
+    using assms
+  proof(induction \<alpha>s arbitrary: s\<^sub>0 M\<^sub>0 )
+    case Nil
+    then show ?case by (auto simp add: close_world_def compose_action_def sound_system_def)
+  next
+    case ass: (Cons \<alpha> \<alpha>s)
+    then obtain pre add del where a: "\<alpha> = Ground_Action pre (Effect add del)"
+      using ground_action.exhaust ast_effect.exhaust by metis
+    let ?M\<^sub>1 = "execute_ground_action \<alpha> M\<^sub>0"
+    have "close_world M\<^sub>0 \<TTurnstile> precondition \<alpha>"
+      using ass(4)
+      by auto
+    moreover have s0_ent_cwM0: "\<forall>fmla\<in>(close_world M\<^sub>0). close_eq s\<^sub>0 \<Turnstile> fmla"
+      using ass(2)
+      unfolding sound_system_def
+      by auto
+    ultimately have s0_ent_alpha_precond: "close_eq s\<^sub>0 \<Turnstile> precondition \<alpha>"
+      unfolding entailment_def
+      by auto
+    then obtain s\<^sub>1 where s1: "(f \<alpha>) s\<^sub>0 = Some s\<^sub>1"
+      "(\<forall>atm. is_predAtom atm \<longrightarrow> atm \<notin> set(dels (effect \<alpha>))
+                                            \<longrightarrow> close_eq s\<^sub>0 \<Turnstile> atm
+                                            \<longrightarrow> close_eq s\<^sub>1 \<Turnstile> atm)"
+      "(\<forall>fmla. fmla \<in> set(adds (effect \<alpha>))
+                                            \<longrightarrow> close_eq s\<^sub>1 \<Turnstile> fmla)"
+      "(\<forall>atm. is_predAtom atm \<and> atm \<notin> set (adds (effect \<alpha>)) \<and> close_eq s\<^sub>0 \<Turnstile> Not atm \<longrightarrow> close_eq s\<^sub>1 \<Turnstile> Not atm)"
+      "(\<forall>fmla. fmla \<in> set (dels (effect \<alpha>)) \<and> fmla \<notin> set(adds (effect \<alpha>)) \<longrightarrow> close_eq s\<^sub>1 \<Turnstile> (Not fmla))"
+      "(\<forall>a b. close_eq s\<^sub>0 \<Turnstile> Atom (Eq a b) \<longrightarrow> close_eq s\<^sub>1 \<Turnstile> Atom (Eq a b))"
+      "(\<forall>a b. close_eq s\<^sub>0 \<Turnstile> Not (Atom (Eq a b)) \<longrightarrow> close_eq s\<^sub>1 \<Turnstile> Not (Atom (Eq a b)))"
+      using ass(2-4)
+      unfolding sound_system_def sound_opr_alt by force
+    have "close_eq s\<^sub>1 \<Turnstile> fmla" if "fmla\<in>close_world ?M\<^sub>1" for fmla
+      using ass(2)
+      using that s1 s0_ent_cwM0
+      unfolding sound_system_def execute_ground_action_def wm_basic_def
+      apply (auto simp: in_close_world_conv)
+      subgoal
+        by (metis (no_types, lifting) DiffE UnE a apply_effect.simps ground_action.sel(2) ast_effect.sel(1) ast_effect.sel(2) close_world_extensive subsetCE)
+      subgoal
+        by (metis Diff_iff Un_iff a ground_action.sel(2) ast_domain.apply_effect.simps ast_domain.close_eq_predAtm ast_effect.sel(1) ast_effect.sel(2) formula_semantics.simps(1) formula_semantics.simps(3) in_close_world_conv is_predAtom.simps(1))
+      done
+    moreover have "(\<forall>atm. fmla \<noteq> formula.Atom atm) \<longrightarrow> s \<Turnstile> fmla" if "fmla\<in>?M\<^sub>1" for fmla s
+    proof-
+      have alpha: "(\<forall>s.\<forall>fmla\<in>set(adds (effect \<alpha>)). \<not> is_predAtom fmla \<longrightarrow> s \<Turnstile> fmla)"
+        using ass(2,3)
+        unfolding sound_system_def ast_domain.sound_opr_alt
+        by auto
+      then show ?thesis
+        using that
+        unfolding a execute_ground_action_def
+        using ass.prems(1)[unfolded sound_system_def]
+        by(cases fmla; fastforce simp: wm_basic_def)
+
+    qed
+    moreover have "(\<forall>opr\<in>\<Sigma>. sound_opr opr (f opr))"
+      using ass(2) unfolding sound_system_def
+      by (auto simp add:)
+    moreover have "wm_basic ?M\<^sub>1"
+      using ass(2,3)
+      unfolding sound_system_def execute_ground_action_def
+      thm sound_opr.cases
+      apply (cases "(\<alpha>,f \<alpha>)" rule: sound_opr.cases)
+      apply (auto simp: wm_basic_def)
+      done
+    ultimately have "sound_system \<Sigma> ?M\<^sub>1 s\<^sub>1 f"
+      unfolding sound_system_def
+      by (auto simp: wm_basic_def)
+    from ass.IH[OF this] ass.prems obtain s' where
+      "compose_actions (map f \<alpha>s) s\<^sub>1 = Some s' \<and> (\<forall>a\<in>close_world M'. s' \<Turnstile>\<^sub>= a)"
+      by auto
+    thus ?case by (auto simp: s1(1))
+  qed
+  with that show ?thesis by blast
+qed
+
+  text \<open>More compact notation of the soundness theorem.\<close>
+  theorem STRIPS_sema_sound_compact_version:
+    "sound_system \<Sigma> M\<^sub>0 s\<^sub>0 f \<Longrightarrow> set \<alpha>s \<subseteq> \<Sigma>
+    \<Longrightarrow> ground_action_path M\<^sub>0 \<alpha>s M'
+    \<Longrightarrow> \<exists>s'. compose_actions (map f \<alpha>s) s\<^sub>0 = Some s'
+          \<and> (\<forall>fmla\<in>close_world M'. s' \<Turnstile>\<^sub>= fmla)"
+    using STRIPS_sema_sound by metis
+
+end \<comment> \<open>Context of \<open>ast_domain\<close>\<close>
+
+subsection \<open>Soundness Theorem for PDDL\<close>
+
+context wf_ast_problem begin
+
+  text \<open>Mapping world models to states\<close>
+  definition state_to_wm :: "state \<Rightarrow> world_model"
+    where "state_to_wm s = ({formula.Atom (predAtm p xs) | p xs. s (p,xs)})"
+  definition wm_to_state :: "world_model \<Rightarrow> state"
+    where "wm_to_state M = (\<lambda>(p,xs). (formula.Atom (predAtm p xs)) \<in> M)"
+
+
+  lemma wm_to_state_eq[simp]: "wm_to_state M (p, as) \<longleftrightarrow> Atom (predAtm p as) \<in> M"
+    by (auto simp: wm_to_state_def)
+
+
+
+
+  lemma wm_to_state_inv[simp]: "wm_to_state (state_to_wm s) = s"
+    by (auto simp: wm_to_state_def
+      state_to_wm_def image_def)
+
+  text \<open>Mapping AST action instances to actions\<close>
+  definition "pddl_opr_to_act g_opr s = (
+    let M = state_to_wm s in
+    if (wm_to_state (close_world M)) \<Turnstile>\<^sub>= (precondition g_opr) then
+      Some (wm_to_state (apply_effect (effect g_opr) M))
+    else
+      None)"
+
+definition "close_eq_M M = (M \<inter> {Atom (predAtm p xs) | p xs. True }) \<union> {Atom (Eq a a) | a. True} \<union> {\<^bold>\<not>(Atom (Eq a b)) | a b. a\<noteq>b}"
+
+  lemma atom_in_wm_eq:
+    "s \<Turnstile>\<^sub>= (formula.Atom atm)
+      \<longleftrightarrow> ((formula.Atom atm) \<in> close_eq_M (state_to_wm s))"
+    by (auto simp: wm_to_state_def
+      state_to_wm_def image_def close_eq_M_def close_eq_def split: atom.splits)
+
+lemma atom_in_wm_2_eq:
+    "close_eq (wm_to_state M) \<Turnstile> (formula.Atom atm)
+      \<longleftrightarrow> ((formula.Atom atm) \<in> close_eq_M M)"
+    by (auto simp: wm_to_state_def
+      state_to_wm_def image_def close_eq_def close_eq_M_def split:atom.splits)
+
+  lemma not_dels_preserved:
+    assumes "f \<notin> (set d)" " f \<in> M"
+    shows "f \<in> apply_effect (Effect a d) M"
+    using assms
+    by auto
+
+  lemma adds_satisfied:
+    assumes "f \<in> (set a)"
+    shows "f \<in> apply_effect (Effect a d) M"
+    using assms
+    by auto
+
+  lemma dels_unsatisfied:
+    assumes "f \<in> (set d)" "f \<notin> set a"
+    shows "f \<notin> apply_effect (Effect a d) M"
+    using assms
+    by auto
+
+  lemma dels_unsatisfied_2:
+    assumes "f \<in> set (dels eff)" "f \<notin> set (adds eff)"
+    shows "f \<notin> apply_effect eff M"
+    using assms
+    by (cases eff; auto)
+
+  lemma wf_fmla_atm_is_atom: "wf_fmla_atom objT f \<Longrightarrow> is_predAtom f"
+    by (cases f rule: wf_fmla_atom.cases) auto
+
+  lemma wf_act_adds_are_atoms:
+    assumes "wf_effect_inst effs" "ae \<in> set (adds effs)"
+    shows "is_predAtom ae"
+    using assms
+    by (cases effs) (auto simp: wf_fmla_atom_alt)
+
+  lemma wf_act_adds_dels_atoms:
+    assumes "wf_effect_inst effs" "ae \<in> set (dels effs)"
+    shows "is_predAtom ae"
+    using assms
+    by (cases effs) (auto simp: wf_fmla_atom_alt)
+
+  lemma to_state_close_from_state_eq[simp]: "wm_to_state (close_world (state_to_wm s)) = s"
+    by (auto simp: wm_to_state_def close_world_def
+      state_to_wm_def image_def)
+
+
+
+lemma wf_eff_pddl_ground_act_is_sound_opr:
+  assumes "wf_effect_inst (effect g_opr)"
+  shows "sound_opr g_opr ((pddl_opr_to_act g_opr))"
+  unfolding sound_opr_alt
+  apply(cases g_opr; safe)
+  subgoal for pre eff s
+    apply (rule exI[where x="wm_to_state(apply_effect eff (state_to_wm s))"])
+    apply (auto simp: pddl_opr_to_act_def Let_def split:if_splits)
+    subgoal for atm
+      by (cases eff; cases atm; auto simp: close_eq_def wm_to_state_def state_to_wm_def split: atom.splits)
+    subgoal for atm
+      by (cases eff; cases atm; auto simp: close_eq_def wm_to_state_def state_to_wm_def split: atom.splits)
+    subgoal for atm
+      using assms
+      by (cases eff; cases atm; force simp: close_eq_def wm_to_state_def state_to_wm_def split: atom.splits)
+    subgoal for fmla
+      using assms
+      by (cases eff; cases fmla rule: wf_fmla_atom.cases; force simp: close_eq_def wm_to_state_def state_to_wm_def split: atom.splits)
+    done
+  subgoal for pre eff fmla
+    using assms
+    by (cases eff; cases fmla rule: wf_fmla_atom.cases; force)
+  done
+
+
+
+  lemma wf_eff_impt_wf_eff_inst: "wf_effect objT eff \<Longrightarrow> wf_effect_inst eff"
+    by (cases eff; auto simp add: wf_fmla_atom_alt)
+
+  lemma wf_pddl_ground_act_is_sound_opr:
+    assumes "wf_ground_action g_opr"
+    shows "sound_opr g_opr (pddl_opr_to_act g_opr)"
+    using wf_eff_impt_wf_eff_inst wf_eff_pddl_ground_act_is_sound_opr assms
+    by (cases g_opr; auto)
+
+  lemma wf_action_schema_sound_inst:
+    assumes "action_params_match act args" "wf_action_schema act"
+    shows "sound_opr
+      (instantiate_action_schema act args)
+      ((pddl_opr_to_act (instantiate_action_schema act args)))"
+    using
+      wf_pddl_ground_act_is_sound_opr[
+        OF wf_instantiate_action_schema[OF assms]]
+      by blast
+
+  lemma wf_plan_act_is_sound:
+    assumes "wf_plan_action (PAction n args)"
+    shows "sound_opr
+      (instantiate_action_schema (the (resolve_action_schema n)) args)
+      ((pddl_opr_to_act
+        (instantiate_action_schema (the (resolve_action_schema n)) args)))"
+    using assms
+    using wf_action_schema_sound_inst wf_eff_pddl_ground_act_is_sound_opr
+    by (auto split: option.splits)
+
+  lemma wf_plan_act_is_sound':
+    assumes "wf_plan_action \<pi>"
+    shows "sound_opr
+      (resolve_instantiate \<pi>)
+      ((pddl_opr_to_act (resolve_instantiate \<pi>)))"
+    using assms wf_plan_act_is_sound
+    by (cases \<pi>; auto )
+
+  lemma wf_world_model_has_atoms: "f\<in>M \<Longrightarrow> wf_world_model M \<Longrightarrow> is_predAtom f"
+    using wf_fmla_atm_is_atom
+    unfolding wf_world_model_def
+    by auto
+
+  lemma wm_to_state_works_for_wf_wm_closed:
+    "wf_world_model M \<Longrightarrow> fmla\<in>close_world M \<Longrightarrow> close_eq (wm_to_state M) \<Turnstile> fmla"
+    apply (cases fmla rule: wf_fmla_atom.cases)
+    by (auto simp: wf_world_model_def close_eq_def wm_to_state_def close_world_def)
+
+  lemma wm_to_state_works_for_wf_wm: "wf_world_model M \<Longrightarrow> fmla\<in>M \<Longrightarrow> close_eq (wm_to_state M) \<Turnstile> fmla"
+    apply (cases fmla rule: wf_fmla_atom.cases)
+    by (auto simp: wf_world_model_def close_eq_def wm_to_state_def)
+
+
+
+  lemma wm_to_state_works_for_I_closed:
+    assumes "x \<in> close_world I"
+    shows "close_eq (wm_to_state I) \<Turnstile> x"
+    apply (rule wm_to_state_works_for_wf_wm_closed)
+    using assms wf_I by auto
+
+
+  lemma wf_wm_imp_basic: "wf_world_model M \<Longrightarrow> wm_basic M"
+    by (auto simp: wf_world_model_def wm_basic_def wf_fmla_atm_is_atom)
+
+theorem wf_plan_sound_system:
+  assumes "\<forall>\<pi>\<in> set \<pi>s. wf_plan_action \<pi>"
+  shows "sound_system
+      (set (map resolve_instantiate \<pi>s))
+      I
+      (wm_to_state I)
+      ((\<lambda>\<alpha>. pddl_opr_to_act \<alpha>))"
+  unfolding sound_system_def
+proof(intro conjI ballI)
+  show "close_eq(wm_to_state I) \<Turnstile> x" if "x \<in> close_world I" for x
+    using that[unfolded in_close_world_conv]
+      wm_to_state_works_for_I_closed wm_to_state_works_for_wf_wm
+    by (auto simp: wf_I)
+
+  show "wm_basic I" using wf_wm_imp_basic[OF wf_I] .
+
+  show "sound_opr \<alpha> (pddl_opr_to_act \<alpha>)" if "\<alpha> \<in> set (map resolve_instantiate \<pi>s)" for \<alpha>
+    using that
+    using wf_plan_act_is_sound' assms
+    by auto
+qed
+
+theorem wf_plan_soundness_theorem:
+    assumes "plan_action_path I \<pi>s M"
+    defines "\<alpha>s \<equiv> map (pddl_opr_to_act \<circ> resolve_instantiate) \<pi>s"
+    defines "s\<^sub>0 \<equiv> wm_to_state I"
+    shows "\<exists>s'. compose_actions \<alpha>s s\<^sub>0 = Some s' \<and> (\<forall>\<phi>\<in>close_world M. s' \<Turnstile>\<^sub>= \<phi>)"
+    apply (rule STRIPS_sema_sound)
+    apply (rule wf_plan_sound_system)
+    using assms
+    unfolding plan_action_path_def
+    by (auto simp add: image_def)
+
+end \<comment> \<open>Context of \<open>wf_ast_problem\<close>\<close>
+
+end

formal/afp/AI_Planning_Languages_Semantics/Option_Monad_Add.thy

@@ -0,0 +1,101 @@
+theory Option_Monad_Add
+imports "HOL-Library.Monad_Syntax"
+begin
+  definition "oassert \<Phi> \<equiv> if \<Phi> then Some () else None"
+
+  fun omap :: "('a\<rightharpoonup>'b) \<Rightarrow> 'a list \<rightharpoonup> 'b list" where
+    "omap f [] = Some []" 
+  | "omap f (x#xs) = do { y \<leftarrow> f x; ys \<leftarrow> omap f xs; Some (y#ys) }"  
+    
+  lemma omap_cong[fundef_cong]:
+    assumes "\<And>x. x\<in>set l' \<Longrightarrow> f x = f' x"
+    assumes "l=l'"
+    shows "omap f l = omap f' l'"
+    unfolding assms(2) using assms(1) by (induction l') (auto)
+
+  lemma assert_eq_iff[simp]: 
+    "oassert \<Phi> = None \<longleftrightarrow> \<not>\<Phi>"  
+    "oassert \<Phi> = Some u \<longleftrightarrow> \<Phi>"  
+    unfolding oassert_def by auto
+
+  lemma omap_length[simp]: "omap f l = Some l' \<Longrightarrow> length l' = length l" 
+    apply (induction l arbitrary: l') 
+    apply (auto split: Option.bind_splits)
+    done 
+
+  lemma omap_append[simp]: "omap f (xs@ys) = do {xs \<leftarrow> omap f xs; ys \<leftarrow> omap f ys; Some (xs@ys)}"
+    by (induction xs) (auto)
+    
+        
+  lemma omap_alt: "omap f l = Some l' \<longleftrightarrow> (l' = map (the o f) l \<and> (\<forall>x\<in>set l. f x \<noteq> None))"  
+    apply (induction l arbitrary: l')
+    apply (auto split: Option.bind_splits)
+    done
+    
+  lemma omap_alt_None: "omap f l = None \<longleftrightarrow> (\<exists>x\<in>set l. f x = None)"
+    apply (induction l)
+    apply (auto split: Option.bind_splits)
+    done
+    
+  lemma omap_nth: "\<lbrakk>omap f l = Some l'; i<length l\<rbrakk> \<Longrightarrow> f (l!i) = Some (l'!i)"
+    apply (induction l arbitrary: l' i)
+    apply (auto split: Option.bind_splits simp: nth_Cons split: nat.splits)
+    done
+
+  lemma omap_eq_Nil_conv[simp]: "omap f xs = Some [] \<longleftrightarrow> xs=[]"
+    apply (cases xs) 
+    apply (auto split: Option.bind_splits)
+    done
+
+  lemma omap_eq_Cons_conv[simp]: "omap f xs = Some (y#ys') \<longleftrightarrow> (\<exists>x xs'. xs=x#xs' \<and> f x = Some y \<and> omap f xs' = Some ys')"  
+    apply (cases xs) 
+    apply (auto split: Option.bind_splits)
+    done
+        
+  lemma omap_eq_append_conv[simp]: "omap f xs = Some (ys\<^sub>1@ys\<^sub>2) \<longleftrightarrow> (\<exists>xs\<^sub>1 xs\<^sub>2. xs=xs\<^sub>1@xs\<^sub>2 \<and> omap f xs\<^sub>1 = Some ys\<^sub>1 \<and> omap f xs\<^sub>2 = Some ys\<^sub>2)"
+    apply (induction ys\<^sub>1 arbitrary: xs)
+    apply (auto 0 3 split: Option.bind_splits)
+    apply (metis append_Cons)
+    done
+  
+  lemma omap_list_all2_conv: "omap f xs = Some ys \<longleftrightarrow> (list_all2 (\<lambda>x y. f x = Some y)) xs ys"  
+    apply (induction xs arbitrary: ys)
+    apply (auto split: Option.bind_splits simp: )
+    apply (simp add: list_all2_Cons1)
+    apply (simp add: list_all2_Cons1)
+    apply (simp add: list_all2_Cons1)
+    apply clarsimp
+    by (metis option.inject)
+    
+    
+    
+    
+  fun omap_option where
+    "omap_option f None = Some None"    
+  | "omap_option f (Some x) = do { x \<leftarrow> f x; Some (Some x) }"
+  
+  lemma omap_option_conv:
+    "omap_option f xx = None \<longleftrightarrow> (\<exists>x. xx=Some x \<and> f x = None)" 
+    "omap_option f xx = (Some (Some x')) \<longleftrightarrow> (\<exists>x. xx=Some x \<and> f x = Some x')"
+    "omap_option f xx = (Some None) \<longleftrightarrow> xx=None"
+    by (cases xx;auto split: Option.bind_splits)+
+  
+  lemma omap_option_eq: "omap_option f x = (case x of None \<Rightarrow> Some None | Some x \<Rightarrow> do { x \<leftarrow> f x; Some (Some x) })"  
+    by (auto split: option.split)
+      
+  fun omap_prod where
+    "omap_prod f\<^sub>1 f\<^sub>2 (a,b) = do { a\<leftarrow>f\<^sub>1 a; b\<leftarrow>f\<^sub>2 b; Some (a,b) }"
+    
+      
+  (* Extend map function for datatype to option monad.
+    TODO: Show reasonable lemmas, like parametricity, etc. 
+    Hopefully only depending on BNF-property of datatype
+   *)
+  definition "omap_dt setf mapf f obj \<equiv> do {
+    oassert (\<forall>x\<in>setf obj. f x \<noteq> None);
+    Some (mapf (the o f) obj)
+  }"
+    
+    
+    
+end

formal/afp/AI_Planning_Languages_Semantics/PDDL_STRIPS_Checker.thy

@@ -0,0 +1,406 @@
+section \<open>Executable PDDL Checker\<close>
+theory PDDL_STRIPS_Checker
+imports
+  PDDL_STRIPS_Semantics
+
+  Error_Monad_Add
+  "HOL.String"
+
+  (*"HOL-Library.Code_Char"     TODO: This might lead to performance loss! CHECK! *)
+  "HOL-Library.Code_Target_Nat"
+
+  "HOL-Library.While_Combinator"
+
+  "Containers.Containers"
+begin
+
+subsection \<open>Generic DFS Reachability Checker\<close>
+text \<open>Used for subtype checks\<close>
+
+definition "E_of_succ succ \<equiv> { (u,v). v\<in>set (succ u) }"
+lemma succ_as_E: "set (succ x) = E_of_succ succ `` {x}"
+  unfolding E_of_succ_def by auto
+
+context
+  fixes succ :: "'a \<Rightarrow> 'a list"
+begin
+
+  private abbreviation (input) "E \<equiv> E_of_succ succ"
+
+
+definition "dfs_reachable D w \<equiv>
+  let (V,w,brk) = while (\<lambda>(V,w,brk). \<not>brk \<and> w\<noteq>[]) (\<lambda>(V,w,_).
+    case w of v#w \<Rightarrow>
+    if D v then (V,v#w,True)
+    else if v\<in>V then (V,w,False)
+    else
+      let V = insert v V in
+      let w = succ v @ w in
+      (V,w,False)
+    ) ({},w,False)
+  in brk"
+
+
+context
+  fixes w\<^sub>0 :: "'a list"
+  assumes finite_dfs_reachable[simp, intro!]: "finite (E\<^sup>* `` set w\<^sub>0)"
+begin
+
+  private abbreviation (input) "W\<^sub>0 \<equiv> set w\<^sub>0"
+
+definition "dfs_reachable_invar D V W brk \<longleftrightarrow>
+    W\<^sub>0 \<subseteq> W \<union> V
+  \<and> W \<union> V \<subseteq> E\<^sup>* `` W\<^sub>0
+  \<and> E``V \<subseteq> W \<union> V
+  \<and> Collect D \<inter> V = {}
+  \<and> (brk \<longrightarrow> Collect D \<inter> E\<^sup>* `` W\<^sub>0 \<noteq> {})"
+
+lemma card_decreases: "
+   \<lbrakk>finite V; y \<notin> V; dfs_reachable_invar D V (Set.insert y W) brk \<rbrakk>
+   \<Longrightarrow> card (E\<^sup>* `` W\<^sub>0 - Set.insert y V) < card (E\<^sup>* `` W\<^sub>0 - V)"
+  apply (rule psubset_card_mono)
+  apply (auto simp: dfs_reachable_invar_def)
+  done
+
+lemma all_neq_Cons_is_Nil[simp]: (* Odd term remaining in goal \<dots> *)
+  "(\<forall>y ys. x2 \<noteq> y # ys) \<longleftrightarrow> x2 = []" by (cases x2) auto
+
+lemma dfs_reachable_correct: "dfs_reachable D w\<^sub>0 \<longleftrightarrow> Collect D \<inter> E\<^sup>* `` set w\<^sub>0 \<noteq> {}"
+  unfolding dfs_reachable_def
+  apply (rule while_rule[where
+    P="\<lambda>(V,w,brk). dfs_reachable_invar D V (set w) brk \<and> finite V"
+    and r="measure (\<lambda>V. card (E\<^sup>* `` (set w\<^sub>0) - V)) <*lex*> measure length <*lex*> measure (\<lambda>True\<Rightarrow>0 | False\<Rightarrow>1)"
+    ])
+  subgoal by (auto simp: dfs_reachable_invar_def)
+  subgoal
+    apply (auto simp: neq_Nil_conv succ_as_E[of succ] split: if_splits)
+    by (auto simp: dfs_reachable_invar_def Image_iff intro: rtrancl.rtrancl_into_rtrancl)
+  subgoal by (fastforce simp: dfs_reachable_invar_def dest: Image_closed_trancl)
+  subgoal by blast
+  subgoal by (auto simp: neq_Nil_conv card_decreases)
+  done
+
+end
+
+definition "tab_succ l \<equiv> Mapping.lookup_default [] (fold (\<lambda>(u,v). Mapping.map_default u [] (Cons v)) l Mapping.empty)"
+
+
+lemma Some_eq_map_option [iff]: "(Some y = map_option f xo) = (\<exists>z. xo = Some z \<and> f z = y)"
+  by (auto simp add: map_option_case split: option.split)
+
+
+lemma tab_succ_correct: "E_of_succ (tab_succ l) = set l"
+proof -
+  have "set (Mapping.lookup_default [] (fold (\<lambda>(u,v). Mapping.map_default u [] (Cons v)) l m) u) = set l `` {u} \<union> set (Mapping.lookup_default [] m u)"
+    for m u
+    apply (induction l arbitrary: m)
+    by (auto
+      simp: Mapping.lookup_default_def Mapping.map_default_def Mapping.default_def
+      simp: lookup_map_entry' lookup_update' keys_is_none_rep Option.is_none_def
+      split: if_splits
+    )
+  from this[where m=Mapping.empty] show ?thesis
+    by (auto simp: E_of_succ_def tab_succ_def lookup_default_empty)
+qed
+
+end
+
+lemma finite_imp_finite_dfs_reachable:
+  "\<lbrakk>finite E; finite S\<rbrakk> \<Longrightarrow> finite (E\<^sup>*``S)"
+  apply (rule finite_subset[where B="S \<union> (Relation.Domain E \<union> Relation.Range E)"])
+  apply (auto simp: intro: finite_Domain finite_Range elim: rtranclE)
+  done
+
+lemma dfs_reachable_tab_succ_correct: "dfs_reachable (tab_succ l) D vs\<^sub>0 \<longleftrightarrow> Collect D \<inter> (set l)\<^sup>*``set vs\<^sub>0 \<noteq> {}"
+  apply (subst dfs_reachable_correct)
+  by (simp_all add: tab_succ_correct finite_imp_finite_dfs_reachable)
+
+
+
+subsection \<open>Implementation Refinements\<close>
+
+subsubsection \<open>Of-Type\<close>
+
+definition "of_type_impl G oT T \<equiv> (\<forall>pt\<in>set (primitives oT). dfs_reachable G ((=) pt) (primitives T))"
+
+
+fun ty_term' where
+  "ty_term' varT objT (term.VAR v) = varT v"
+| "ty_term' varT objT (term.CONST c) = Mapping.lookup objT c"
+
+lemma ty_term'_correct_aux: "ty_term' varT objT t = ty_term varT (Mapping.lookup objT) t"
+  by (cases t) auto
+
+lemma ty_term'_correct[simp]: "ty_term' varT objT = ty_term varT (Mapping.lookup objT)"
+  using ty_term'_correct_aux by auto
+
+context ast_domain begin
+
+  definition "of_type1 pt T \<longleftrightarrow> pt \<in> subtype_rel\<^sup>* `` set (primitives T)"
+
+  lemma of_type_refine1: "of_type oT T \<longleftrightarrow> (\<forall>pt\<in>set (primitives oT). of_type1 pt T)"
+    unfolding of_type_def of_type1_def by auto
+
+  definition "STG \<equiv> (tab_succ (map subtype_edge (types D)))"
+
+  lemma subtype_rel_impl: "subtype_rel = E_of_succ (tab_succ (map subtype_edge (types D)))"
+    by (simp add: tab_succ_correct subtype_rel_def)
+
+  lemma of_type1_impl: "of_type1 pt T \<longleftrightarrow> dfs_reachable (tab_succ (map subtype_edge (types D))) ((=)pt) (primitives T)"
+    by (simp add: subtype_rel_impl of_type1_def dfs_reachable_tab_succ_correct tab_succ_correct)
+
+  lemma of_type_impl_correct: "of_type_impl STG oT T \<longleftrightarrow> of_type oT T"
+    unfolding of_type1_impl STG_def of_type_impl_def of_type_refine1 ..
+
+  definition mp_constT :: "(object, type) mapping" where
+    "mp_constT = Mapping.of_alist (consts D)"
+
+  lemma mp_objT_correct[simp]: "Mapping.lookup mp_constT = constT"
+    unfolding mp_constT_def constT_def
+    by transfer (simp add: Map_To_Mapping.map_apply_def)
+
+
+
+
+
+
+  text \<open>Lifting the subtype-graph through wf-checker\<close>
+  context
+    fixes ty_ent :: "'ent \<rightharpoonup> type"  \<comment> \<open>Entity's type, None if invalid\<close>
+  begin
+
+    definition "is_of_type' stg v T \<longleftrightarrow> (
+      case ty_ent v of
+        Some vT \<Rightarrow> of_type_impl stg vT T
+      | None \<Rightarrow> False)"
+
+    lemma is_of_type'_correct: "is_of_type' STG v T = is_of_type ty_ent v T"
+      unfolding is_of_type'_def is_of_type_def of_type_impl_correct ..
+
+    fun wf_pred_atom' where "wf_pred_atom' stg (p,vs) \<longleftrightarrow> (case sig p of
+          None \<Rightarrow> False
+        | Some Ts \<Rightarrow> list_all2 (is_of_type' stg) vs Ts)"
+
+    lemma wf_pred_atom'_correct: "wf_pred_atom' STG pvs = wf_pred_atom ty_ent pvs"
+      by (cases pvs) (auto simp: is_of_type'_correct[abs_def] split:option.split)
+
+    fun wf_atom' :: "_ \<Rightarrow> 'ent atom \<Rightarrow> bool" where
+      "wf_atom' stg (atom.predAtm p vs) \<longleftrightarrow> wf_pred_atom' stg (p,vs)"
+    | "wf_atom' stg (atom.Eq a b) = (ty_ent a \<noteq> None \<and> ty_ent b \<noteq> None)"
+
+    lemma wf_atom'_correct: "wf_atom' STG a = wf_atom ty_ent a"
+      by (cases a) (auto simp: wf_pred_atom'_correct is_of_type'_correct[abs_def] split: option.splits)
+
+    fun wf_fmla' :: "_ \<Rightarrow> ('ent atom) formula \<Rightarrow> bool" where
+      "wf_fmla' stg (Atom a) \<longleftrightarrow> wf_atom' stg a"
+    | "wf_fmla' stg \<bottom> \<longleftrightarrow> True"
+    | "wf_fmla' stg (\<phi>1 \<^bold>\<and> \<phi>2) \<longleftrightarrow> (wf_fmla' stg \<phi>1 \<and> wf_fmla' stg \<phi>2)"
+    | "wf_fmla' stg (\<phi>1 \<^bold>\<or> \<phi>2) \<longleftrightarrow> (wf_fmla' stg \<phi>1 \<and> wf_fmla' stg \<phi>2)"
+    | "wf_fmla' stg (\<phi>1 \<^bold>\<rightarrow> \<phi>2) \<longleftrightarrow> (wf_fmla' stg \<phi>1 \<and> wf_fmla' stg \<phi>2)"
+    | "wf_fmla' stg (\<^bold>\<not>\<phi>) \<longleftrightarrow> wf_fmla' stg \<phi>"
+
+    lemma wf_fmla'_correct: "wf_fmla' STG \<phi> \<longleftrightarrow> wf_fmla ty_ent \<phi>"
+      by (induction \<phi> rule: wf_fmla.induct) (auto simp: wf_atom'_correct)
+
+    fun wf_fmla_atom1' where
+      "wf_fmla_atom1' stg (Atom (predAtm p vs)) \<longleftrightarrow> wf_pred_atom' stg (p,vs)"
+    | "wf_fmla_atom1' stg _ \<longleftrightarrow> False"
+
+    lemma wf_fmla_atom1'_correct: "wf_fmla_atom1' STG \<phi> = wf_fmla_atom ty_ent \<phi>"
+      by (cases \<phi> rule: wf_fmla_atom.cases) (auto
+        simp: wf_atom'_correct is_of_type'_correct[abs_def] split: option.splits)
+
+    fun wf_effect' where
+      "wf_effect' stg (Effect a d) \<longleftrightarrow>
+          (\<forall>ae\<in>set a. wf_fmla_atom1' stg ae)
+        \<and> (\<forall>de\<in>set d.  wf_fmla_atom1' stg de)"
+
+    lemma wf_effect'_correct: "wf_effect' STG e = wf_effect ty_ent e"
+      by (cases e) (auto simp: wf_fmla_atom1'_correct)
+
+  end \<comment> \<open>Context fixing \<open>ty_ent\<close>\<close>
+
+  fun wf_action_schema' :: "_ \<Rightarrow> _ \<Rightarrow> ast_action_schema \<Rightarrow> bool" where
+    "wf_action_schema' stg conT (Action_Schema n params pre eff) \<longleftrightarrow> (
+      let
+        tyv = ty_term' (map_of params) conT
+      in
+        distinct (map fst params)
+      \<and> wf_fmla' tyv stg pre
+      \<and> wf_effect' tyv stg eff)"
+
+  lemma wf_action_schema'_correct: "wf_action_schema' STG mp_constT s = wf_action_schema s"
+    by (cases s) (auto simp: wf_fmla'_correct wf_effect'_correct)
+
+  definition wf_domain' :: "_ \<Rightarrow> _ \<Rightarrow> bool" where
+    "wf_domain' stg conT \<equiv>
+      wf_types
+    \<and> distinct (map (predicate_decl.pred) (predicates D))
+    \<and> (\<forall>p\<in>set (predicates D). wf_predicate_decl p)
+    \<and> distinct (map fst (consts D))
+    \<and> (\<forall>(n,T)\<in>set (consts D). wf_type T)
+    \<and> distinct (map ast_action_schema.name (actions D))
+    \<and> (\<forall>a\<in>set (actions D). wf_action_schema' stg conT a)
+    "
+
+  lemma wf_domain'_correct: "wf_domain' STG mp_constT = wf_domain"
+    unfolding wf_domain_def wf_domain'_def
+    by (auto simp: wf_action_schema'_correct)
+
+
+end \<comment> \<open>Context of \<open>ast_domain\<close>\<close>
+
+subsubsection \<open>Application of Effects\<close>
+
+context ast_domain begin
+  text \<open>We implement the application of an effect by explicit iteration over
+    the additions and deletions\<close>
+  fun apply_effect_exec
+    :: "object ast_effect \<Rightarrow> world_model \<Rightarrow> world_model"
+  where
+    "apply_effect_exec (Effect a d) s
+      = fold (\<lambda>add s. Set.insert add s) a
+          (fold (\<lambda>del s. Set.remove del s) d s)"
+
+  lemma apply_effect_exec_refine[simp]:
+    "apply_effect_exec (Effect (a) (d)) s
+    = apply_effect (Effect (a) (d)) s"
+  proof(induction a arbitrary: s)
+    case Nil
+    then show ?case
+    proof(induction d arbitrary: s)
+      case Nil
+      then show ?case by auto
+    next
+      case (Cons a d)
+      then show ?case
+        by (auto simp add: image_def)
+    qed
+  next
+    case (Cons a a)
+    then show ?case
+    proof(induction d arbitrary: s)
+      case Nil
+      then show ?case by (auto; metis Set.insert_def sup_assoc insert_iff)
+    next
+      case (Cons a d)
+      then show ?case
+        by (auto simp: Un_commute minus_set_fold union_set_fold)
+    qed
+  qed
+
+  lemmas apply_effect_eq_impl_eq
+    = apply_effect_exec_refine[symmetric, unfolded apply_effect_exec.simps]
+
+end \<comment> \<open>Context of \<open>ast_domain\<close>\<close>
+
+subsubsection \<open>Well-Formedness\<close>
+
+context ast_problem begin
+
+  text \<open> We start by defining a mapping from objects to types. The container
+    framework will generate efficient, red-black tree based code for that
+    later. \<close>
+
+  type_synonym objT = "(object, type) mapping"
+
+  definition mp_objT :: "(object, type) mapping" where
+    "mp_objT = Mapping.of_alist (consts D @ objects P)"
+
+  lemma mp_objT_correct[simp]: "Mapping.lookup mp_objT = objT"
+    unfolding mp_objT_def objT_alt
+    by transfer (simp add: Map_To_Mapping.map_apply_def)
+
+  text \<open>We refine the typecheck to use the mapping\<close>
+
+  definition "is_obj_of_type_impl stg mp n T = (
+    case Mapping.lookup mp n of None \<Rightarrow> False | Some oT \<Rightarrow> of_type_impl stg oT T
+  )"
+
+  lemma is_obj_of_type_impl_correct[simp]:
+    "is_obj_of_type_impl STG mp_objT = is_obj_of_type"
+    apply (intro ext)
+    apply (auto simp: is_obj_of_type_impl_def is_obj_of_type_def of_type_impl_correct split: option.split)
+    done
+
+  text \<open>We refine the well-formedness checks to use the mapping\<close>
+
+  definition wf_fact' :: "objT \<Rightarrow> _ \<Rightarrow> fact \<Rightarrow> bool"
+    where
+    "wf_fact' ot stg \<equiv> wf_pred_atom' (Mapping.lookup ot) stg"
+
+  lemma wf_fact'_correct[simp]: "wf_fact' mp_objT STG = wf_fact"
+    by (auto simp: wf_fact'_def wf_fact_def wf_pred_atom'_correct[abs_def])
+
+
+  definition "wf_fmla_atom2' mp stg f
+    = (case f of formula.Atom (predAtm p vs) \<Rightarrow> (wf_fact' mp stg (p,vs)) | _ \<Rightarrow> False)"
+
+  lemma wf_fmla_atom2'_correct[simp]:
+    "wf_fmla_atom2' mp_objT STG \<phi> = wf_fmla_atom objT \<phi>"
+    apply (cases \<phi> rule: wf_fmla_atom.cases)
+    by (auto simp: wf_fmla_atom2'_def wf_fact_def split: option.splits)
+
+  definition "wf_problem' stg conT mp \<equiv>
+      wf_domain' stg conT
+    \<and> distinct (map fst (objects P) @ map fst (consts D))
+    \<and> (\<forall>(n,T)\<in>set (objects P). wf_type T)
+    \<and> distinct (init P)
+    \<and> (\<forall>f\<in>set (init P). wf_fmla_atom2' mp stg f)
+    \<and> wf_fmla' (Mapping.lookup mp) stg (goal P)"
+
+  lemma wf_problem'_correct:
+    "wf_problem' STG mp_constT mp_objT = wf_problem"
+    unfolding wf_problem_def wf_problem'_def wf_world_model_def
+    by (auto simp: wf_domain'_correct wf_fmla'_correct)
+
+
+  text \<open>Instantiating actions will yield well-founded effects.
+    Corollary of @{thm wf_instantiate_action_schema}.\<close>
+  lemma wf_effect_inst_weak:
+    fixes a args
+    defines "ai \<equiv> instantiate_action_schema a args"
+    assumes A: "action_params_match a args"
+      "wf_action_schema a"
+    shows "wf_effect_inst (effect ai)"
+    using wf_instantiate_action_schema[OF A] unfolding ai_def[symmetric]
+    by (cases ai) (auto simp: wf_effect_inst_alt)
+
+
+end \<comment> \<open>Context of \<open>ast_problem\<close>\<close>
+
+
+context wf_ast_domain begin
+  text \<open>Resolving an action yields a well-founded action schema.\<close>
+  (* TODO: This must be implicitly proved when showing that plan execution
+    preserves wf. Try to remove this redundancy!*)
+  lemma resolve_action_wf:
+    assumes "resolve_action_schema n = Some a"
+    shows "wf_action_schema a"
+  proof -
+    from wf_domain have
+      X1: "distinct (map ast_action_schema.name (actions D))"
+      and X2: "\<forall>a\<in>set (actions D). wf_action_schema a"
+      unfolding wf_domain_def by auto
+
+    show ?thesis
+      using assms unfolding resolve_action_schema_def
+      by (auto simp add: index_by_eq_Some_eq[OF X1] X2)
+  qed
+
+end \<comment> \<open>Context of \<open>ast_domain\<close>\<close>
+
+
+subsubsection \<open>Execution of Plan Actions\<close>
+
+text \<open>We will perform two refinement steps, to summarize redundant operations\<close>
+
+text \<open>We first lift action schema lookup into the error monad. \<close>
+context ast_domain begin
+  definition "resolve_action_schemaE n \<equiv>
+    lift_opt
+      (resolve_action_schema n)
+      (ERR (shows ''No such action schema '' o shows n))"
+end \<comment> \<open>Context of \<open>ast_domain\<close>\<close>
+
+end \<comment> \<open>Theory\<close>

formal/afp/AI_Planning_Languages_Semantics/PDDL_STRIPS_Semantics.thy

@@ -0,0 +1,969 @@
+section \<open>PDDL and STRIPS Semantics\<close>
+theory PDDL_STRIPS_Semantics
+imports
+  "Propositional_Proof_Systems.Formulas"
+  "Propositional_Proof_Systems.Sema"
+  "Propositional_Proof_Systems.Consistency"
+  "Automatic_Refinement.Misc"
+  "Automatic_Refinement.Refine_Util"
+begin
+no_notation insert ("_ \<triangleright> _" [56,55] 55)
+
+subsection \<open>Utility Functions\<close>
+definition "index_by f l \<equiv> map_of (map (\<lambda>x. (f x,x)) l)"
+
+lemma index_by_eq_Some_eq[simp]:
+  assumes "distinct (map f l)"
+  shows "index_by f l n = Some x \<longleftrightarrow> (x\<in>set l \<and> f x = n)"
+  unfolding index_by_def
+  using assms
+  by (auto simp: o_def)
+
+lemma index_by_eq_SomeD:
+  shows "index_by f l n = Some x \<Longrightarrow> (x\<in>set l \<and> f x = n)"
+  unfolding index_by_def
+  by (auto dest: map_of_SomeD)
+
+
+lemma lookup_zip_idx_eq:
+  assumes "length params = length args"
+  assumes "i<length args"
+  assumes "distinct params"
+  assumes "k = params ! i"
+  shows "map_of (zip params args) k = Some (args ! i)"
+  using assms
+  by (auto simp: in_set_conv_nth)
+
+lemma rtrancl_image_idem[simp]: "R\<^sup>* `` R\<^sup>* `` s = R\<^sup>* `` s"
+  by (metis relcomp_Image rtrancl_idemp_self_comp)
+
+
+subsection \<open>Abstract Syntax\<close>
+
+subsubsection \<open>Generic Entities\<close>
+type_synonym name = string
+
+datatype predicate = Pred (name: name)
+
+text \<open>Some of the AST entities are defined over a polymorphic \<open>'val\<close> type,
+  which gets either instantiated by variables (for domains)
+  or objects (for problems).
+\<close>
+
+text \<open>An atom is either a predicate with arguments, or an equality statement.\<close>
+datatype 'ent atom = predAtm (predicate: predicate) (arguments: "'ent list")
+                     | Eq (lhs: 'ent) (rhs: 'ent)
+
+text \<open>A type is a list of primitive type names.
+  To model a primitive type, we use a singleton list.\<close>
+datatype type = Either (primitives: "name list")
+
+text \<open>An effect contains a list of values to be added, and a list of values
+  to be removed.\<close>
+datatype 'ent ast_effect = Effect (adds: "('ent atom formula) list") (dels: "('ent atom formula) list")
+
+text \<open>Variables are identified by their names.\<close>
+datatype variable = varname: Var name
+text \<open>Objects and constants are identified by their names\<close>
+datatype object = name: Obj name
+
+datatype "term" = VAR variable | CONST object
+hide_const (open) VAR CONST \<comment> \<open>Refer to constructors by qualified names only\<close>
+
+
+
+
+subsubsection \<open>Domains\<close>
+
+text \<open>An action schema has a name, a typed parameter list, a precondition,
+  and an effect.\<close>
+datatype ast_action_schema = Action_Schema
+  (name: name)
+  (parameters: "(variable \<times> type) list")
+  (precondition: "term atom formula")
+  (effect: "term ast_effect")
+
+text \<open>A predicate declaration contains the predicate's name and its
+  argument types.\<close>
+datatype predicate_decl = PredDecl
+  (pred: predicate)
+  (argTs: "type list")
+
+text \<open>A domain contains the declarations of primitive types, predicates,
+  and action schemas.\<close>
+datatype ast_domain = Domain
+  (types: "(name \<times> name) list") \<comment> \<open> \<open>(type, supertype)\<close> declarations. \<close>
+  (predicates: "predicate_decl list")
+  ("consts": "(object \<times> type) list")
+  (actions: "ast_action_schema list")
+
+subsubsection \<open>Problems\<close>
+
+
+text \<open>A fact is a predicate applied to objects.\<close>
+type_synonym fact = "predicate \<times> object list"
+
+text \<open>A problem consists of a domain, a list of objects,
+  a description of the initial state, and a description of the goal state. \<close>
+datatype ast_problem = Problem
+  (domain: ast_domain)
+  (objects: "(object \<times> type) list")
+  (init: "object atom formula list")
+  (goal: "object atom formula")
+
+
+subsubsection \<open>Plans\<close>
+datatype plan_action = PAction
+  (name: name)
+  (arguments: "object list")
+
+type_synonym plan = "plan_action list"
+
+subsubsection \<open>Ground Actions\<close>
+text \<open>The following datatype represents an action scheme that has been
+  instantiated by replacing the arguments with concrete objects,
+  also called ground action.
+\<close>
+datatype ground_action = Ground_Action
+  (precondition: "(object atom) formula")
+  (effect: "object ast_effect")
+
+
+
+subsection \<open>Closed-World Assumption, Equality, and Negation\<close>
+  text \<open>Discriminator for atomic predicate formulas.\<close>
+  fun is_predAtom where
+    "is_predAtom (Atom (predAtm _ _)) = True" | "is_predAtom _ = False"
+
+
+  text \<open>The world model is a set of (atomic) formulas\<close>
+  type_synonym world_model = "object atom formula set"
+
+  text \<open>It is basic, if it only contains atoms\<close>
+  definition "wm_basic M \<equiv> \<forall>a\<in>M. is_predAtom a"
+
+  text \<open>A valuation extracted from the atoms of the world model\<close>
+  definition valuation :: "world_model \<Rightarrow> object atom valuation"
+    where "valuation M \<equiv> \<lambda>predAtm p xs \<Rightarrow> Atom (predAtm p xs) \<in> M | Eq a b \<Rightarrow> a=b"
+
+  text \<open>Augment a world model by adding negated versions of all atoms
+    not contained in it, as well as interpretations of equality.\<close>
+  definition close_world :: "world_model \<Rightarrow> world_model" where "close_world M =
+    M \<union> {\<^bold>\<not>(Atom (predAtm p as)) | p as. Atom (predAtm p as) \<notin> M}
+    \<union> {Atom (Eq a a) | a. True} \<union> {\<^bold>\<not>(Atom (Eq a b)) | a b. a\<noteq>b}"
+
+  definition "close_neg M \<equiv> M \<union> {\<^bold>\<not>(Atom a) | a. Atom a \<notin> M}"
+  lemma "wm_basic M \<Longrightarrow> close_world M = close_neg (M \<union> {Atom (Eq a a) | a. True})"
+    unfolding close_world_def close_neg_def wm_basic_def
+    apply clarsimp
+    apply (auto 0 3)
+    by (metis atom.exhaust)
+
+
+  abbreviation cw_entailment (infix "\<^sup>c\<TTurnstile>\<^sub>=" 53) where
+    "M \<^sup>c\<TTurnstile>\<^sub>= \<phi> \<equiv> close_world M \<TTurnstile> \<phi>"
+
+
+  lemma
+    close_world_extensive: "M \<subseteq> close_world M" and
+    close_world_idem[simp]: "close_world (close_world M) = close_world M"
+    by (auto simp: close_world_def)
+
+  lemma in_close_world_conv:
+    "\<phi> \<in> close_world M \<longleftrightarrow> (
+        \<phi>\<in>M
+      \<or> (\<exists>p as. \<phi>=\<^bold>\<not>(Atom (predAtm p as)) \<and> Atom (predAtm p as)\<notin>M)
+      \<or> (\<exists>a. \<phi>=Atom (Eq a a))
+      \<or> (\<exists>a b. \<phi>=\<^bold>\<not>(Atom (Eq a b)) \<and> a\<noteq>b)
+    )"
+    by (auto simp: close_world_def)
+
+  lemma valuation_aux_1:
+    fixes M :: world_model and \<phi> :: "object atom formula"
+    defines "C \<equiv> close_world M"
+    assumes A: "\<forall>\<phi>\<in>C. \<A> \<Turnstile> \<phi>"
+    shows "\<A> = valuation M"
+    using A unfolding C_def
+    apply -
+    apply (auto simp: in_close_world_conv valuation_def Ball_def intro!: ext split: atom.split)
+    apply (metis formula_semantics.simps(1) formula_semantics.simps(3))
+    apply (metis formula_semantics.simps(1) formula_semantics.simps(3))
+    by (metis atom.collapse(2) formula_semantics.simps(1) is_predAtm_def)
+
+
+
+  lemma valuation_aux_2:
+    assumes "wm_basic M"
+    shows "(\<forall>G\<in>close_world M. valuation M \<Turnstile> G)"
+    using assms unfolding wm_basic_def
+    by (force simp: in_close_world_conv valuation_def elim: is_predAtom.elims)
+
+  lemma val_imp_close_world: "valuation M \<Turnstile> \<phi> \<Longrightarrow> M \<^sup>c\<TTurnstile>\<^sub>= \<phi>"
+    unfolding entailment_def
+    using valuation_aux_1
+    by blast
+
+  lemma close_world_imp_val:
+    "wm_basic M \<Longrightarrow> M \<^sup>c\<TTurnstile>\<^sub>= \<phi> \<Longrightarrow> valuation M \<Turnstile> \<phi>"
+    unfolding entailment_def using valuation_aux_2 by blast
+
+  text \<open>Main theorem of this section:
+    If a world model \<open>M\<close> contains only atoms, its induced valuation
+    satisfies a formula \<open>\<phi>\<close> if and only if the closure of \<open>M\<close> entails \<open>\<phi>\<close>.
+
+    Note that there are no syntactic restrictions on \<open>\<phi>\<close>,
+    in particular, \<open>\<phi>\<close> may contain negation.
+  \<close>
+  theorem valuation_iff_close_world:
+    assumes "wm_basic M"
+    shows "valuation M \<Turnstile> \<phi> \<longleftrightarrow> M \<^sup>c\<TTurnstile>\<^sub>= \<phi>"
+    using assms val_imp_close_world close_world_imp_val by blast
+
+
+subsubsection \<open>Proper Generalization\<close>
+text \<open>Adding negation and equality is a proper generalization of the
+  case without negation and equality\<close>
+
+fun is_STRIPS_fmla :: "'ent atom formula \<Rightarrow> bool" where
+  "is_STRIPS_fmla (Atom (predAtm _ _)) \<longleftrightarrow> True"
+| "is_STRIPS_fmla (\<bottom>) \<longleftrightarrow> True"
+| "is_STRIPS_fmla (\<phi>\<^sub>1 \<^bold>\<and> \<phi>\<^sub>2) \<longleftrightarrow> is_STRIPS_fmla \<phi>\<^sub>1 \<and> is_STRIPS_fmla \<phi>\<^sub>2"
+| "is_STRIPS_fmla (\<phi>\<^sub>1 \<^bold>\<or> \<phi>\<^sub>2) \<longleftrightarrow> is_STRIPS_fmla \<phi>\<^sub>1 \<and> is_STRIPS_fmla \<phi>\<^sub>2"
+| "is_STRIPS_fmla (\<^bold>\<not>\<bottom>) \<longleftrightarrow> True"
+| "is_STRIPS_fmla _ \<longleftrightarrow> False"
+
+lemma aux1: "\<lbrakk>wm_basic M; is_STRIPS_fmla \<phi>; valuation M \<Turnstile> \<phi>; \<forall>G\<in>M. \<A> \<Turnstile> G\<rbrakk> \<Longrightarrow> \<A> \<Turnstile> \<phi>"
+  apply(induction \<phi> rule: is_STRIPS_fmla.induct)
+  by (auto simp: valuation_def)
+
+lemma aux2: "\<lbrakk>wm_basic M; is_STRIPS_fmla \<phi>; \<forall>\<A>. (\<forall>G\<in>M. \<A> \<Turnstile> G) \<longrightarrow> \<A> \<Turnstile> \<phi>\<rbrakk> \<Longrightarrow> valuation M \<Turnstile> \<phi>"
+  apply(induction \<phi> rule: is_STRIPS_fmla.induct)
+  apply simp_all
+  apply (metis in_close_world_conv valuation_aux_2)
+  using in_close_world_conv valuation_aux_2 apply blast
+  using in_close_world_conv valuation_aux_2 by auto
+
+
+lemma valuation_iff_STRIPS:
+  assumes "wm_basic M"
+  assumes "is_STRIPS_fmla \<phi>"
+  shows "valuation M \<Turnstile> \<phi> \<longleftrightarrow> M \<TTurnstile> \<phi>"
+proof -
+  have aux1: "\<And>\<A>. \<lbrakk>valuation M \<Turnstile> \<phi>; \<forall>G\<in>M. \<A> \<Turnstile> G\<rbrakk> \<Longrightarrow> \<A> \<Turnstile> \<phi>"
+    using assms apply(induction \<phi> rule: is_STRIPS_fmla.induct)
+    by (auto simp: valuation_def)
+  have aux2: "\<lbrakk>\<forall>\<A>. (\<forall>G\<in>M. \<A> \<Turnstile> G) \<longrightarrow> \<A> \<Turnstile> \<phi>\<rbrakk> \<Longrightarrow> valuation M \<Turnstile> \<phi>"
+    using assms
+    apply(induction \<phi> rule: is_STRIPS_fmla.induct)
+    apply simp_all
+    apply (metis in_close_world_conv valuation_aux_2)
+    using in_close_world_conv valuation_aux_2 apply blast
+    using in_close_world_conv valuation_aux_2 by auto
+  show ?thesis
+    by (auto simp: entailment_def intro: aux1 aux2)
+qed
+
+text \<open>Our extension to negation and equality is a proper generalization of the
+  standard STRIPS semantics for formula without negation and equality\<close>
+theorem proper_STRIPS_generalization:
+  "\<lbrakk>wm_basic M; is_STRIPS_fmla \<phi>\<rbrakk> \<Longrightarrow> M \<^sup>c\<TTurnstile>\<^sub>= \<phi> \<longleftrightarrow> M \<TTurnstile> \<phi>"
+  by (simp add: valuation_iff_close_world[symmetric] valuation_iff_STRIPS)
+
+subsection \<open>STRIPS Semantics\<close>
+
+text \<open>For this section, we fix a domain \<open>D\<close>, using Isabelle's
+  locale mechanism.\<close>
+locale ast_domain =
+  fixes D :: ast_domain
+begin
+  text \<open>It seems to be agreed upon that, in case of a contradictory effect,
+    addition overrides deletion. We model this behaviour by first executing
+    the deletions, and then the additions.\<close>
+  fun apply_effect :: "object ast_effect \<Rightarrow> world_model \<Rightarrow> world_model"
+  where
+     "apply_effect (Effect a d) s = (s - set d) \<union> (set a)"
+
+  text \<open>Execute a ground action\<close>
+  definition execute_ground_action :: "ground_action \<Rightarrow> world_model \<Rightarrow> world_model"
+  where
+    "execute_ground_action a M = apply_effect (effect a) M"
+
+  text \<open>Predicate to model that the given list of action instances is
+    executable, and transforms an initial world model \<open>M\<close> into a final
+    model \<open>M'\<close>.
+
+    Note that this definition over the list structure is more convenient in HOL
+    than to explicitly define an indexed sequence \<open>M\<^sub>0\<dots>M\<^sub>N\<close> of intermediate world
+     models, as done in [Lif87].
+  \<close>
+  fun ground_action_path
+    :: "world_model \<Rightarrow> ground_action list \<Rightarrow> world_model \<Rightarrow> bool"
+  where
+    "ground_action_path M [] M' \<longleftrightarrow> (M = M')"
+  | "ground_action_path M (\<alpha>#\<alpha>s) M' \<longleftrightarrow> M \<^sup>c\<TTurnstile>\<^sub>= precondition \<alpha>
+    \<and> ground_action_path (execute_ground_action \<alpha> M) \<alpha>s M'"
+
+  text \<open>Function equations as presented in paper,
+    with inlined @{const execute_ground_action}.\<close>
+  lemma ground_action_path_in_paper:
+    "ground_action_path M [] M' \<longleftrightarrow> (M = M')"
+    "ground_action_path M (\<alpha>#\<alpha>s) M' \<longleftrightarrow> M \<^sup>c\<TTurnstile>\<^sub>= precondition \<alpha>
+    \<and> (ground_action_path (apply_effect (effect \<alpha>) M) \<alpha>s M')"
+    by (auto simp: execute_ground_action_def)
+
+end \<comment> \<open>Context of \<open>ast_domain\<close>\<close>
+
+
+
+subsection \<open>Well-Formedness of PDDL\<close>
+
+(* Well-formedness *)
+
+(*
+  Compute signature: predicate/arity
+  Check that all atoms (schemas and facts) satisfy signature
+
+  for action:
+    Check that used parameters \<subseteq> declared parameters
+
+  for init/goal: Check that facts only use declared objects
+*)
+
+
+fun ty_term where
+  "ty_term varT objT (term.VAR v) = varT v"
+| "ty_term varT objT (term.CONST c) = objT c"
+
+
+lemma ty_term_mono: "varT \<subseteq>\<^sub>m varT' \<Longrightarrow> objT \<subseteq>\<^sub>m objT' \<Longrightarrow>
+  ty_term varT objT \<subseteq>\<^sub>m ty_term varT' objT'"
+  apply (rule map_leI)
+  subgoal for x v
+    apply (cases x)
+    apply (auto dest: map_leD)
+    done
+  done
+
+
+context ast_domain begin
+
+  text \<open>The signature is a partial function that maps the predicates
+    of the domain to lists of argument types.\<close>
+  definition sig :: "predicate \<rightharpoonup> type list" where
+    "sig \<equiv> map_of (map (\<lambda>PredDecl p n \<Rightarrow> (p,n)) (predicates D))"
+
+  text \<open>We use a flat subtype hierarchy, where every type is a subtype
+    of object, and there are no other subtype relations.
+
+    Note that we do not need to restrict this relation to declared types,
+    as we will explicitly ensure that all types used in the problem are
+    declared.
+    \<close>
+
+  fun subtype_edge where
+    "subtype_edge (ty,superty) = (superty,ty)"
+
+  definition "subtype_rel \<equiv> set (map subtype_edge (types D))"
+
+  (*
+  definition "subtype_rel \<equiv> {''object''}\<times>UNIV"
+  *)
+
+  definition of_type :: "type \<Rightarrow> type \<Rightarrow> bool" where
+    "of_type oT T \<equiv> set (primitives oT) \<subseteq> subtype_rel\<^sup>* `` set (primitives T)"
+  text \<open>This checks that every primitive on the LHS is contained in or a
+    subtype of a primitive on the RHS\<close>
+
+
+  text \<open>For the next few definitions, we fix a partial function that maps
+    a polymorphic entity type @{typ "'e"} to types. An entity can be
+    instantiated by variables or objects later.\<close>
+  context
+    fixes ty_ent :: "'ent \<rightharpoonup> type"  \<comment> \<open>Entity's type, None if invalid\<close>
+  begin
+
+    text \<open>Checks whether an entity has a given type\<close>
+    definition is_of_type :: "'ent \<Rightarrow> type \<Rightarrow> bool" where
+      "is_of_type v T \<longleftrightarrow> (
+        case ty_ent v of
+          Some vT \<Rightarrow> of_type vT T
+        | None \<Rightarrow> False)"
+
+    fun wf_pred_atom :: "predicate \<times> 'ent list \<Rightarrow> bool" where
+      "wf_pred_atom (p,vs) \<longleftrightarrow> (
+        case sig p of
+          None \<Rightarrow> False
+        | Some Ts \<Rightarrow> list_all2 is_of_type vs Ts)"
+
+    text \<open>Predicate-atoms are well-formed if their arguments match the
+      signature, equalities are well-formed if the arguments are valid
+      objects (have a type).
+
+      TODO: We could check that types may actually overlap
+    \<close>
+    fun wf_atom :: "'ent atom \<Rightarrow> bool" where
+      "wf_atom (predAtm p vs) \<longleftrightarrow> wf_pred_atom (p,vs)"
+    | "wf_atom (Eq a b) \<longleftrightarrow> ty_ent a \<noteq> None \<and> ty_ent b \<noteq> None"
+
+    text \<open>A formula is well-formed if it consists of valid atoms,
+      and does not contain negations, except for the encoding \<open>\<^bold>\<not>\<bottom>\<close> of true.
+    \<close>
+    fun wf_fmla :: "('ent atom) formula \<Rightarrow> bool" where
+      "wf_fmla (Atom a) \<longleftrightarrow> wf_atom a"
+    | "wf_fmla (\<bottom>) \<longleftrightarrow> True"
+    | "wf_fmla (\<phi>1 \<^bold>\<and> \<phi>2) \<longleftrightarrow> (wf_fmla \<phi>1 \<and> wf_fmla \<phi>2)"
+    | "wf_fmla (\<phi>1 \<^bold>\<or> \<phi>2) \<longleftrightarrow> (wf_fmla \<phi>1 \<and> wf_fmla \<phi>2)"
+    | "wf_fmla (\<^bold>\<not>\<phi>) \<longleftrightarrow> wf_fmla \<phi>"
+    | "wf_fmla (\<phi>1 \<^bold>\<rightarrow> \<phi>2) \<longleftrightarrow> (wf_fmla \<phi>1 \<and> wf_fmla \<phi>2)"
+
+    lemma "wf_fmla \<phi> = (\<forall>a\<in>atoms \<phi>. wf_atom a)"
+      by (induction \<phi>) auto
+
+    (*lemma wf_fmla_add_simps[simp]: "wf_fmla (\<^bold>\<not>\<phi>) \<longleftrightarrow> \<phi>=\<bottom>"
+      by (cases \<phi>) auto*)
+
+    text \<open>Special case for a well-formed atomic predicate formula\<close>
+    fun wf_fmla_atom where
+      "wf_fmla_atom (Atom (predAtm a vs)) \<longleftrightarrow> wf_pred_atom (a,vs)"
+    | "wf_fmla_atom _ \<longleftrightarrow> False"
+
+    lemma wf_fmla_atom_alt: "wf_fmla_atom \<phi> \<longleftrightarrow> is_predAtom \<phi> \<and> wf_fmla \<phi>"
+      by (cases \<phi> rule: wf_fmla_atom.cases) auto
+
+    text \<open>An effect is well-formed if the added and removed formulas
+      are atomic\<close>
+    fun wf_effect where
+      "wf_effect (Effect a d) \<longleftrightarrow>
+          (\<forall>ae\<in>set a. wf_fmla_atom ae)
+        \<and> (\<forall>de\<in>set d.  wf_fmla_atom de)"
+
+  end \<comment> \<open>Context fixing \<open>ty_ent\<close>\<close>
+
+
+  definition constT :: "object \<rightharpoonup> type" where
+    "constT \<equiv> map_of (consts D)"
+
+  text \<open>An action schema is well-formed if the parameter names are distinct,
+    and the precondition and effect is well-formed wrt.\ the parameters.
+  \<close>
+  fun wf_action_schema :: "ast_action_schema \<Rightarrow> bool" where
+    "wf_action_schema (Action_Schema n params pre eff) \<longleftrightarrow> (
+      let
+        tyt = ty_term (map_of params) constT
+      in
+        distinct (map fst params)
+      \<and> wf_fmla tyt pre
+      \<and> wf_effect tyt eff)"
+
+  text \<open>A type is well-formed if it consists only of declared primitive types,
+     and the type object.\<close>
+  fun wf_type where
+    "wf_type (Either Ts) \<longleftrightarrow> set Ts \<subseteq> insert ''object'' (fst`set (types D))"
+
+  text \<open>A predicate is well-formed if its argument types are well-formed.\<close>
+  fun wf_predicate_decl where
+    "wf_predicate_decl (PredDecl p Ts) \<longleftrightarrow> (\<forall>T\<in>set Ts. wf_type T)"
+
+  text \<open>The types declaration is well-formed, if all supertypes are declared types (or object)\<close>
+  definition "wf_types \<equiv> snd`set (types D) \<subseteq> insert ''object'' (fst`set (types D))"
+
+  text \<open>A domain is well-formed if
+    \<^item> there are no duplicate declared predicate names,
+    \<^item> all declared predicates are well-formed,
+    \<^item> there are no duplicate action names,
+    \<^item> and all declared actions are well-formed
+    \<close>
+  definition wf_domain :: "bool" where
+    "wf_domain \<equiv>
+      wf_types
+    \<and> distinct (map (predicate_decl.pred) (predicates D))
+    \<and> (\<forall>p\<in>set (predicates D). wf_predicate_decl p)
+    \<and> distinct (map fst (consts D))
+    \<and> (\<forall>(n,T)\<in>set (consts D). wf_type T)
+    \<and> distinct (map ast_action_schema.name (actions D))
+    \<and> (\<forall>a\<in>set (actions D). wf_action_schema a)
+    "
+
+end \<comment> \<open>locale \<open>ast_domain\<close>\<close>
+
+text \<open>We fix a problem, and also include the definitions for the domain
+  of this problem.\<close>
+locale ast_problem = ast_domain "domain P"
+  for P :: ast_problem
+begin
+  text \<open>We refer to the problem domain as \<open>D\<close>\<close>
+  abbreviation "D \<equiv> ast_problem.domain P"
+
+  definition objT :: "object \<rightharpoonup> type" where
+    "objT \<equiv> map_of (objects P) ++ constT"
+
+  lemma objT_alt: "objT = map_of (consts D @ objects P)"
+    unfolding objT_def constT_def
+    apply (clarsimp)
+    done
+
+  definition wf_fact :: "fact \<Rightarrow> bool" where
+    "wf_fact = wf_pred_atom objT"
+
+  text \<open>This definition is needed for well-formedness of the initial model,
+    and forward-references to the concept of world model.
+  \<close>
+  definition wf_world_model where
+    "wf_world_model M = (\<forall>f\<in>M. wf_fmla_atom objT f)"
+
+  (*Note: current semantics assigns each object a unique type *)
+  definition wf_problem where
+    "wf_problem \<equiv>
+      wf_domain
+    \<and> distinct (map fst (objects P) @ map fst (consts D))
+    \<and> (\<forall>(n,T)\<in>set (objects P). wf_type T)
+    \<and> distinct (init P)
+    \<and> wf_world_model (set (init P))
+    \<and> wf_fmla objT (goal P)
+    "
+
+  fun wf_effect_inst :: "object ast_effect \<Rightarrow> bool" where
+    "wf_effect_inst (Effect (a) (d))
+      \<longleftrightarrow> (\<forall>a\<in>set a \<union> set d. wf_fmla_atom objT a)"
+
+  lemma wf_effect_inst_alt: "wf_effect_inst eff = wf_effect objT eff"
+    by (cases eff) auto
+
+end \<comment> \<open>locale \<open>ast_problem\<close>\<close>
+
+text \<open>Locale to express a well-formed domain\<close>
+locale wf_ast_domain = ast_domain +
+  assumes wf_domain: wf_domain
+
+text \<open>Locale to express a well-formed problem\<close>
+locale wf_ast_problem = ast_problem P for P +
+  assumes wf_problem: wf_problem
+begin
+  sublocale wf_ast_domain "domain P"
+    apply unfold_locales
+    using wf_problem
+    unfolding wf_problem_def by simp
+
+end \<comment> \<open>locale \<open>wf_ast_problem\<close>\<close>
+
+subsection \<open>PDDL Semantics\<close>
+
+(* Semantics *)
+
+(*  To apply plan_action:
+    find action schema, instantiate, check precond, apply effect
+*)
+
+
+
+context ast_domain begin
+
+  definition resolve_action_schema :: "name \<rightharpoonup> ast_action_schema" where
+    "resolve_action_schema n = index_by ast_action_schema.name (actions D) n"
+
+  fun subst_term where
+    "subst_term psubst (term.VAR x) = psubst x"
+  | "subst_term psubst (term.CONST c) = c"
+
+  text \<open>To instantiate an action schema, we first compute a substitution from
+    parameters to objects, and then apply this substitution to the
+    precondition and effect. The substitution is applied via the \<open>map_xxx\<close>
+    functions generated by the datatype package.
+    \<close>
+  fun instantiate_action_schema
+    :: "ast_action_schema \<Rightarrow> object list \<Rightarrow> ground_action"
+  where
+    "instantiate_action_schema (Action_Schema n params pre eff) args = (let
+        tsubst = subst_term (the o (map_of (zip (map fst params) args)));
+        pre_inst = (map_formula o map_atom) tsubst pre;
+        eff_inst = (map_ast_effect) tsubst eff
+      in
+        Ground_Action pre_inst eff_inst
+      )"
+
+end \<comment> \<open>Context of \<open>ast_domain\<close>\<close>
+
+
+context ast_problem begin
+
+  text \<open>Initial model\<close>
+  definition I :: "world_model" where
+    "I \<equiv> set (init P)"
+
+
+  text \<open>Resolve a plan action and instantiate the referenced action schema.\<close>
+  fun resolve_instantiate :: "plan_action \<Rightarrow> ground_action" where
+    "resolve_instantiate (PAction n args) =
+      instantiate_action_schema
+        (the (resolve_action_schema n))
+        args"
+
+  text \<open>Check whether object has specified type\<close>
+  definition "is_obj_of_type n T \<equiv> case objT n of
+    None \<Rightarrow> False
+  | Some oT \<Rightarrow> of_type oT T"
+
+  text \<open>We can also use the generic \<open>is_of_type\<close> function.\<close>
+  lemma is_obj_of_type_alt: "is_obj_of_type = is_of_type objT"
+    apply (intro ext)
+    unfolding is_obj_of_type_def is_of_type_def by auto
+
+
+  text \<open>HOL encoding of matching an action's formal parameters against an
+    argument list.
+    The parameters of the action are encoded as a list of \<open>name\<times>type\<close> pairs,
+    such that we map it to a list of types first. Then, the list
+    relator @{const list_all2} checks that arguments and types have the same
+    length, and each matching pair of argument and type
+    satisfies the predicate @{const is_obj_of_type}.
+  \<close>
+  definition "action_params_match a args
+    \<equiv> list_all2 is_obj_of_type args (map snd (parameters a))"
+
+  text \<open>At this point, we can define well-formedness of a plan action:
+    The action must refer to a declared action schema, the arguments must
+    be compatible with the formal parameters' types.
+  \<close>
+ (* Objects are valid and match parameter types *)
+  fun wf_plan_action :: "plan_action \<Rightarrow> bool" where
+    "wf_plan_action (PAction n args) = (
+      case resolve_action_schema n of
+        None \<Rightarrow> False
+      | Some a \<Rightarrow>
+          action_params_match a args
+        \<and> wf_effect_inst (effect (instantiate_action_schema a args))
+        )"
+  text \<open>
+    TODO: The second conjunct is redundant, as instantiating a well formed
+      action with valid objects yield a valid effect.
+  \<close>
+
+
+
+  text \<open>A sequence of plan actions form a path, if they are well-formed and
+    their instantiations form a path.\<close>
+  definition plan_action_path
+    :: "world_model \<Rightarrow> plan_action list \<Rightarrow> world_model \<Rightarrow> bool"
+  where
+    "plan_action_path M \<pi>s M' =
+        ((\<forall>\<pi> \<in> set \<pi>s. wf_plan_action \<pi>)
+      \<and> ground_action_path M (map resolve_instantiate \<pi>s) M')"
+
+  text \<open>A plan is valid wrt.\ a given initial model, if it forms a path to a
+    goal model \<close>
+  definition valid_plan_from :: "world_model \<Rightarrow> plan \<Rightarrow> bool" where
+    "valid_plan_from M \<pi>s = (\<exists>M'. plan_action_path M \<pi>s M' \<and> M' \<^sup>c\<TTurnstile>\<^sub>= (goal P))"
+
+  (* Implementation note: resolve and instantiate already done inside
+      enabledness check, redundancy! *)
+
+  text \<open>Finally, a plan is valid if it is valid wrt.\ the initial world
+    model @{const I}\<close>
+  definition valid_plan :: "plan \<Rightarrow> bool"
+    where "valid_plan \<equiv> valid_plan_from I"
+
+  text \<open>Concise definition used in paper:\<close>
+  lemma "valid_plan \<pi>s \<equiv> \<exists>M'. plan_action_path I \<pi>s M' \<and> M' \<^sup>c\<TTurnstile>\<^sub>= (goal P)"
+    unfolding valid_plan_def valid_plan_from_def by auto
+
+
+end \<comment> \<open>Context of \<open>ast_problem\<close>\<close>
+
+
+
+subsection \<open>Preservation of Well-Formedness\<close>
+
+subsubsection \<open>Well-Formed Action Instances\<close>
+text \<open>The goal of this section is to establish that well-formedness of
+  world models is preserved by execution of well-formed plan actions.
+\<close>
+
+context ast_problem begin
+
+  text \<open>As plan actions are executed by first instantiating them, and then
+    executing the action instance, it is natural to define a well-formedness
+    concept for action instances.\<close>
+
+  fun wf_ground_action :: "ground_action \<Rightarrow> bool" where
+    "wf_ground_action (Ground_Action pre eff) \<longleftrightarrow> (
+        wf_fmla objT pre
+      \<and> wf_effect objT eff
+      )
+    "
+
+  text \<open>We first prove that instantiating a well-formed action schema will yield
+    a well-formed action instance.
+
+    We begin with some auxiliary lemmas before the actual theorem.
+  \<close>
+
+  lemma (in ast_domain) of_type_refl[simp, intro!]: "of_type T T"
+    unfolding of_type_def by auto
+
+  lemma (in ast_domain) of_type_trans[trans]:
+    "of_type T1 T2 \<Longrightarrow> of_type T2 T3 \<Longrightarrow> of_type T1 T3"
+    unfolding of_type_def
+    by clarsimp (metis (no_types, opaque_lifting)
+      Image_mono contra_subsetD order_refl rtrancl_image_idem)
+
+  lemma is_of_type_map_ofE:
+    assumes "is_of_type (map_of params) x T"
+    obtains i xT where "i<length params" "params!i = (x,xT)" "of_type xT T"
+    using assms
+    unfolding is_of_type_def
+    by (auto split: option.splits dest!: map_of_SomeD simp: in_set_conv_nth)
+
+  lemma wf_atom_mono:
+    assumes SS: "tys \<subseteq>\<^sub>m tys'"
+    assumes WF: "wf_atom tys a"
+    shows "wf_atom tys' a"
+  proof -
+    have "list_all2 (is_of_type tys') xs Ts" if "list_all2 (is_of_type tys) xs Ts" for xs Ts
+      using that
+      apply induction
+      by (auto simp: is_of_type_def split: option.splits dest: map_leD[OF SS])
+    with WF show ?thesis
+      by (cases a) (auto split: option.splits dest: map_leD[OF SS])
+  qed
+
+  lemma wf_fmla_atom_mono:
+    assumes SS: "tys \<subseteq>\<^sub>m tys'"
+    assumes WF: "wf_fmla_atom tys a"
+    shows "wf_fmla_atom tys' a"
+  proof -
+    have "list_all2 (is_of_type tys') xs Ts" if "list_all2 (is_of_type tys) xs Ts" for xs Ts
+      using that
+      apply induction
+      by (auto simp: is_of_type_def split: option.splits dest: map_leD[OF SS])
+    with WF show ?thesis
+      by (cases a rule: wf_fmla_atom.cases) (auto split: option.splits dest: map_leD[OF SS])
+  qed
+
+
+  lemma constT_ss_objT: "constT \<subseteq>\<^sub>m objT"
+    unfolding constT_def objT_def
+    apply rule
+    by (auto simp: map_add_def split: option.split)
+
+  lemma wf_atom_constT_imp_objT: "wf_atom (ty_term Q constT) a \<Longrightarrow> wf_atom (ty_term Q objT) a"
+    apply (erule wf_atom_mono[rotated])
+    apply (rule ty_term_mono)
+    by (simp_all add: constT_ss_objT)
+
+  lemma wf_fmla_atom_constT_imp_objT: "wf_fmla_atom (ty_term Q constT) a \<Longrightarrow> wf_fmla_atom (ty_term Q objT) a"
+    apply (erule wf_fmla_atom_mono[rotated])
+    apply (rule ty_term_mono)
+    by (simp_all add: constT_ss_objT)
+
+  context
+    fixes Q and f :: "variable \<Rightarrow> object"
+    assumes INST: "is_of_type Q x T \<Longrightarrow> is_of_type objT (f x) T"
+  begin
+
+    lemma is_of_type_var_conv: "is_of_type (ty_term Q objT) (term.VAR x) T  \<longleftrightarrow> is_of_type Q x T"
+      unfolding is_of_type_def by (auto)
+
+    lemma is_of_type_const_conv: "is_of_type (ty_term Q objT) (term.CONST x) T  \<longleftrightarrow> is_of_type objT x T"
+      unfolding is_of_type_def
+      by (auto split: option.split)
+
+    lemma INST': "is_of_type (ty_term Q objT) x T \<Longrightarrow> is_of_type objT (subst_term f x) T"
+      apply (cases x) using INST apply (auto simp: is_of_type_var_conv is_of_type_const_conv)
+      done
+
+
+    lemma wf_inst_eq_aux: "Q x = Some T \<Longrightarrow> objT (f x) \<noteq> None"
+      using INST[of x T] unfolding is_of_type_def
+      by (auto split: option.splits)
+
+    lemma wf_inst_eq_aux': "ty_term Q objT x = Some T \<Longrightarrow> objT (subst_term f x) \<noteq> None"
+      by (cases x) (auto simp: wf_inst_eq_aux)
+
+
+    lemma wf_inst_atom:
+      assumes "wf_atom (ty_term Q constT) a"
+      shows "wf_atom objT (map_atom (subst_term f) a)"
+    proof -
+      have X1: "list_all2 (is_of_type objT) (map (subst_term f) xs) Ts" if
+        "list_all2 (is_of_type (ty_term Q objT)) xs Ts" for xs Ts
+        using that
+        apply induction
+        using INST'
+        by auto
+      then show ?thesis
+        using assms[THEN wf_atom_constT_imp_objT] wf_inst_eq_aux'
+        by (cases a; auto split: option.splits)
+
+    qed
+
+    lemma wf_inst_formula_atom:
+      assumes "wf_fmla_atom (ty_term Q constT) a"
+      shows "wf_fmla_atom objT ((map_formula o map_atom o subst_term) f a)"
+      using assms[THEN wf_fmla_atom_constT_imp_objT] wf_inst_atom
+      apply (cases a rule: wf_fmla_atom.cases; auto split: option.splits)
+      by (simp add: INST' list.rel_map(1) list_all2_mono)
+
+    lemma wf_inst_effect:
+      assumes "wf_effect (ty_term Q constT) \<phi>"
+      shows "wf_effect objT ((map_ast_effect o subst_term) f \<phi>)"
+      using assms
+      proof (induction \<phi>)
+        case (Effect x1a x2a)
+        then show ?case using wf_inst_formula_atom by auto
+      qed
+
+    lemma wf_inst_formula:
+      assumes "wf_fmla (ty_term Q constT) \<phi>"
+      shows "wf_fmla objT ((map_formula o map_atom o subst_term) f \<phi>)"
+      using assms
+      by (induction \<phi>) (auto simp: wf_inst_atom dest: wf_inst_eq_aux)
+
+  end
+
+
+
+  text \<open>Instantiating a well-formed action schema with compatible arguments
+    will yield a well-formed action instance.
+  \<close>
+  theorem wf_instantiate_action_schema:
+    assumes "action_params_match a args"
+    assumes "wf_action_schema a"
+    shows "wf_ground_action (instantiate_action_schema a args)"
+  proof (cases a)
+    case [simp]: (Action_Schema name params pre eff)
+    have INST:
+      "is_of_type objT ((the \<circ> map_of (zip (map fst params) args)) x) T"
+      if "is_of_type (map_of params) x T" for x T
+      using that
+      apply (rule is_of_type_map_ofE)
+      using assms
+      apply (clarsimp simp: Let_def)
+      subgoal for i xT
+        unfolding action_params_match_def
+        apply (subst lookup_zip_idx_eq[where i=i];
+          (clarsimp simp: list_all2_lengthD)?)
+        apply (frule list_all2_nthD2[where p=i]; simp?)
+        apply (auto
+                simp: is_obj_of_type_alt is_of_type_def
+                intro: of_type_trans
+                split: option.splits)
+        done
+      done
+    then show ?thesis
+      using assms(2) wf_inst_formula wf_inst_effect
+      by (fastforce split: term.splits simp: Let_def comp_apply[abs_def])
+  qed
+end \<comment> \<open>Context of \<open>ast_problem\<close>\<close>
+
+
+
+subsubsection \<open>Preservation\<close>
+
+context ast_problem begin
+
+  text \<open>We start by defining two shorthands for enabledness and execution of
+    a plan action.\<close>
+
+  text \<open>Shorthand for enabled plan action: It is well-formed, and the
+    precondition holds for its instance.\<close>
+  definition plan_action_enabled :: "plan_action \<Rightarrow> world_model \<Rightarrow> bool" where
+    "plan_action_enabled \<pi> M
+      \<longleftrightarrow> wf_plan_action \<pi> \<and> M \<^sup>c\<TTurnstile>\<^sub>= precondition (resolve_instantiate \<pi>)"
+
+  text \<open>Shorthand for executing a plan action: Resolve, instantiate, and
+    apply effect\<close>
+  definition execute_plan_action :: "plan_action \<Rightarrow> world_model \<Rightarrow> world_model"
+    where "execute_plan_action \<pi> M
+      = (apply_effect (effect (resolve_instantiate \<pi>)) M)"
+
+  text \<open>The @{const plan_action_path} predicate can be decomposed naturally
+    using these shorthands: \<close>
+  lemma plan_action_path_Nil[simp]: "plan_action_path M [] M' \<longleftrightarrow> M'=M"
+    by (auto simp: plan_action_path_def)
+
+  lemma plan_action_path_Cons[simp]:
+    "plan_action_path M (\<pi>#\<pi>s) M' \<longleftrightarrow>
+      plan_action_enabled \<pi> M
+    \<and> plan_action_path (execute_plan_action \<pi> M) \<pi>s M'"
+    by (auto
+      simp: plan_action_path_def execute_plan_action_def
+            execute_ground_action_def plan_action_enabled_def)
+
+
+
+end \<comment> \<open>Context of \<open>ast_problem\<close>\<close>
+
+context wf_ast_problem begin
+  text \<open>The initial world model is well-formed\<close>
+  lemma wf_I: "wf_world_model I"
+    using wf_problem
+    unfolding I_def wf_world_model_def wf_problem_def
+    apply(safe) subgoal for f by (induction f) auto
+    done
+
+  text \<open>Application of a well-formed effect preserves well-formedness
+    of the model\<close>
+  lemma wf_apply_effect:
+    assumes "wf_effect objT e"
+    assumes "wf_world_model s"
+    shows "wf_world_model (apply_effect e s)"
+    using assms wf_problem
+    unfolding wf_world_model_def wf_problem_def wf_domain_def
+    by (cases e) (auto split: formula.splits prod.splits)
+
+  text \<open>Execution of plan actions preserves well-formedness\<close>
+  theorem wf_execute:
+    assumes "plan_action_enabled \<pi> s"
+    assumes "wf_world_model s"
+    shows "wf_world_model (execute_plan_action \<pi> s)"
+    using assms
+  proof (cases \<pi>)
+    case [simp]: (PAction name args)
+
+    from \<open>plan_action_enabled \<pi> s\<close> have "wf_plan_action \<pi>"
+      unfolding plan_action_enabled_def by auto
+    then obtain a where
+      "resolve_action_schema name = Some a" and
+      T: "action_params_match a args"
+      by (auto split: option.splits)
+
+    from wf_domain have
+      [simp]: "distinct (map ast_action_schema.name (actions D))"
+      unfolding wf_domain_def by auto
+
+    from \<open>resolve_action_schema name = Some a\<close> have
+      "a \<in> set (actions D)"
+      unfolding resolve_action_schema_def by auto
+    with wf_domain have "wf_action_schema a"
+      unfolding wf_domain_def by auto
+    hence "wf_ground_action (resolve_instantiate \<pi>)"
+      using \<open>resolve_action_schema name = Some a\<close> T
+        wf_instantiate_action_schema
+      by auto
+    thus ?thesis
+      apply (simp add: execute_plan_action_def execute_ground_action_def)
+      apply (rule wf_apply_effect)
+      apply (cases "resolve_instantiate \<pi>"; simp)
+      by (rule \<open>wf_world_model s\<close>)
+  qed
+
+  theorem wf_execute_compact_notation:
+    "plan_action_enabled \<pi> s \<Longrightarrow> wf_world_model s
+    \<Longrightarrow> wf_world_model (execute_plan_action \<pi> s)"
+    by (rule wf_execute)
+
+
+  text \<open>Execution of a plan preserves well-formedness\<close>
+  corollary wf_plan_action_path:
+    assumes "wf_world_model M" and " plan_action_path M \<pi>s M'"
+    shows "wf_world_model M'"
+    using assms
+    by (induction \<pi>s arbitrary: M) (auto intro: wf_execute)
+
+
+end \<comment> \<open>Context of \<open>wf_ast_problem\<close>\<close>
+
+
+
+
+end \<comment> \<open>Theory\<close>

formal/afp/AI_Planning_Languages_Semantics/SASP_Checker.thy

@@ -0,0 +1,348 @@
+theory SASP_Checker
+imports SASP_Semantics
+  "HOL-Library.Code_Target_Nat"
+begin
+
+section \<open>An Executable Checker for Multi-Valued Planning Problem Solutions\<close>
+
+
+  subsection \<open>Auxiliary Lemmas\<close>
+  lemma map_of_leI:
+    assumes "distinct (map fst l)"
+    assumes "\<And>k v. (k,v)\<in>set l \<Longrightarrow> m k = Some v"
+    shows "map_of l \<subseteq>\<^sub>m m"  
+    using assms
+    by (metis (no_types, opaque_lifting) domIff map_le_def map_of_SomeD not_Some_eq)  
+
+  lemma [simp]: "fst \<circ> (\<lambda>(a, b, c, d). (f a b c d, g a b c d)) = (\<lambda>(a,b,c,d). f a b c d)"
+    by auto
+      
+  lemma map_mp: "m\<subseteq>\<^sub>mm' \<Longrightarrow> m k = Some v \<Longrightarrow> m' k = Some v"    
+    by (auto simp: map_le_def dom_def)
+      
+      
+  lemma map_add_map_of_fold: 
+    fixes ps and m :: "'a \<rightharpoonup> 'b"
+    assumes "distinct (map fst ps)"
+    shows "m ++ map_of ps = fold (\<lambda>(k, v) m. m(k \<mapsto> v)) ps m"
+  proof -
+    have X1: "fold (\<lambda>(k, v) m. m(k \<mapsto> v)) ps m(a \<mapsto> b) 
+            = fold (\<lambda>(k, v) m. m(k \<mapsto> v)) ps (m(a \<mapsto> b))" 
+      if "a \<notin> fst ` set ps"
+      for a b ps and m :: "'a \<rightharpoonup> 'b"
+      using that
+      by (induction ps arbitrary: m) (auto simp: fun_upd_twist)
+    
+    show ?thesis
+      using assms
+      by (induction ps arbitrary: m) (auto simp: X1)
+  qed    
+
+  
+
+  subsection \<open>Well-formedness Check\<close>
+  lemmas wf_code_thms = 
+      ast_problem.astDom_def ast_problem.astI_def ast_problem.astG_def ast_problem.ast\<delta>_def
+      ast_problem.numVars_def ast_problem.numVals_def 
+      ast_problem.wf_partial_state_def ast_problem.wf_operator_def ast_problem.well_formed_def
+      
+      
+  declare wf_code_thms[code]
+      
+  export_code ast_problem.well_formed in SML
+
+    
+  subsection \<open>Execution\<close>  
+    
+  definition match_pre :: "ast_precond \<Rightarrow> state \<Rightarrow> bool" where
+    "match_pre \<equiv> \<lambda>(x,v) s. s x = Some v"
+    
+  definition match_pres :: "ast_precond list \<Rightarrow> state \<Rightarrow> bool" where 
+    "match_pres pres s \<equiv> \<forall>pre\<in>set pres. match_pre pre s"
+    
+  definition match_implicit_pres :: "ast_effect list \<Rightarrow> state \<Rightarrow> bool" where
+    "match_implicit_pres effs s \<equiv> \<forall>(_,x,vp,_)\<in>set effs. 
+      (case vp of None \<Rightarrow> True | Some v \<Rightarrow> s x = Some v)"
+    
+  definition enabled_opr' :: "ast_operator \<Rightarrow> state \<Rightarrow> bool" where 
+    "enabled_opr' \<equiv> \<lambda>(name,pres,effs,cost) s. match_pres pres s \<and> match_implicit_pres effs s"
+      
+  definition eff_enabled' :: "state \<Rightarrow> ast_effect \<Rightarrow> bool" where
+    "eff_enabled' s \<equiv> \<lambda>(pres,_,_,_). match_pres pres s"
+    
+  definition "execute_opr' \<equiv> \<lambda>(name,_,effs,_) s. 
+    let effs = filter (eff_enabled' s) effs
+    in fold (\<lambda>(_,x,_,v) s. s(x\<mapsto>v)) effs s
+  "  
+
+  definition lookup_operator' :: "ast_problem \<Rightarrow> name \<rightharpoonup> ast_operator" 
+    where "lookup_operator' \<equiv> \<lambda>(D,I,G,\<delta>) name. find (\<lambda>(n,_,_,_). n=name) \<delta>"
+    
+  definition enabled' :: "ast_problem \<Rightarrow> name \<Rightarrow> state \<Rightarrow> bool" where
+    "enabled' problem name s \<equiv> 
+      case lookup_operator' problem name of 
+        Some \<pi> \<Rightarrow> enabled_opr' \<pi> s
+      | None \<Rightarrow> False"
+
+  definition execute' :: "ast_problem \<Rightarrow> name \<Rightarrow> state \<Rightarrow> state" where
+    "execute' problem name s \<equiv> 
+      case lookup_operator' problem name of 
+        Some \<pi> \<Rightarrow> execute_opr' \<pi> s
+      | None \<Rightarrow> undefined"
+    
+    
+  context wf_ast_problem begin  
+    
+    lemma match_pres_correct:
+      assumes D: "distinct (map fst pres)"
+      assumes "s\<in>valid_states"  
+      shows "match_pres pres s \<longleftrightarrow> s\<in>subsuming_states (map_of pres)"
+    proof -
+      have "match_pres pres s \<longleftrightarrow> map_of pres \<subseteq>\<^sub>m s"
+        unfolding match_pres_def match_pre_def
+        apply (auto split: prod.splits)
+        using map_le_def map_of_SomeD apply fastforce  
+        by (metis (full_types) D domIff map_le_def map_of_eq_Some_iff option.simps(3))
+    
+      with assms show ?thesis          
+        unfolding subsuming_states_def 
+        by auto
+    qed
+       
+    lemma match_implicit_pres_correct:
+      assumes D: "distinct (map (\<lambda>(_, v, _, _). v) effs)"
+      assumes "s\<in>valid_states"  
+      shows "match_implicit_pres effs s \<longleftrightarrow> s\<in>subsuming_states (map_of (implicit_pres effs))"
+    proof -
+      from assms show ?thesis
+        unfolding subsuming_states_def 
+        unfolding match_implicit_pres_def implicit_pres_def
+        apply (auto 
+            split: prod.splits option.splits 
+            simp: distinct_map_filter
+            intro!: map_of_leI)
+        apply (force simp: distinct_map_filter split: prod.split elim: map_mp)  
+        done  
+    qed
+          
+    lemma enabled_opr'_correct:
+      assumes V: "s\<in>valid_states"
+      assumes "lookup_operator name = Some \<pi>"  
+      shows "enabled_opr' \<pi> s \<longleftrightarrow> enabled name s"  
+      using lookup_operator_wf[OF assms(2)] assms
+      unfolding enabled_opr'_def enabled_def wf_operator_def
+      by (auto 
+          simp: match_pres_correct[OF _ V] match_implicit_pres_correct[OF _ V]
+          simp: wf_partial_state_def
+          split: option.split
+          )  
+       
+    lemma eff_enabled'_correct:
+      assumes V: "s\<in>valid_states"
+      assumes "case eff of (pres,_,_,_) \<Rightarrow> wf_partial_state pres"  
+      shows "eff_enabled' s eff \<longleftrightarrow> eff_enabled s eff"  
+      using assms  
+      unfolding eff_enabled'_def eff_enabled_def wf_partial_state_def
+      by (auto simp: match_pres_correct)  
+    
+    
+    lemma execute_opr'_correct:
+      assumes V: "s\<in>valid_states"
+      assumes LO: "lookup_operator name = Some \<pi>"  
+      shows "execute_opr' \<pi> s = execute name s"  
+    proof (cases \<pi>)
+      case [simp]: (fields name pres effs)
+        
+      have [simp]: "filter (eff_enabled' s) effs = filter (eff_enabled s) effs"  
+        apply (rule filter_cong[OF refl])
+        apply (rule eff_enabled'_correct[OF V])
+        using lookup_operator_wf[OF LO]  
+        unfolding wf_operator_def by auto  
+          
+      have X1: "distinct (map fst (map (\<lambda>(_, x, _, y). (x, y)) (filter (eff_enabled s) effs)))"
+        using lookup_operator_wf[OF LO]
+        unfolding wf_operator_def 
+        by (auto simp: distinct_map_filter)
+        
+      term "filter (eff_enabled s) effs"    
+          
+      have [simp]: 
+        "fold (\<lambda>(_, x, _, v) s. s(x \<mapsto> v)) l s =
+         fold (\<lambda>(k, v) m. m(k \<mapsto> v)) (map (\<lambda>(_, x, _, y). (x, y)) l) s" 
+        for l :: "ast_effect list"
+        by (induction l arbitrary: s) auto
+          
+      show ?thesis 
+        unfolding execute_opr'_def execute_def using LO
+        by (auto simp: map_add_map_of_fold[OF X1])
+    qed
+        
+      
+    lemma lookup_operator'_correct: 
+      "lookup_operator' problem name = lookup_operator name"
+      unfolding lookup_operator'_def lookup_operator_def
+      unfolding ast\<delta>_def  
+      by (auto split: prod.split)  
+        
+    lemma enabled'_correct:
+      assumes V: "s\<in>valid_states"
+      shows "enabled' problem name s = enabled name s"
+      unfolding enabled'_def  
+      using enabled_opr'_correct[OF V]
+      by (auto split: option.splits simp: enabled_def lookup_operator'_correct)  
+        
+    lemma execute'_correct:
+      assumes V: "s\<in>valid_states"
+      assumes "enabled name s"      (* Intentionally put this here, also we could resolve non-enabled case by reflexivity (undefined=undefined) *)
+      shows "execute' problem name s = execute name s"  
+      unfolding execute'_def  
+      using execute_opr'_correct[OF V] \<open>enabled name s\<close>
+      by (auto split: option.splits simp: enabled_def lookup_operator'_correct)  
+        
+        
+        
+  end    
+
+  context ast_problem 
+  begin  
+    
+    fun simulate_plan :: "plan \<Rightarrow> state \<rightharpoonup> state" where
+      "simulate_plan [] s = Some s"
+    | "simulate_plan (\<pi>#\<pi>s) s = (
+        if enabled \<pi> s then 
+          let s' = execute \<pi> s in
+          simulate_plan \<pi>s s'
+        else
+          None
+      )"  
+    
+    lemma simulate_plan_correct: "simulate_plan \<pi>s s = Some s' \<longleftrightarrow> path_to s \<pi>s s'"
+      by (induction s \<pi>s s' rule: path_to.induct) auto  
+      
+    definition check_plan :: "plan \<Rightarrow> bool" where
+      "check_plan \<pi>s = (
+        case simulate_plan \<pi>s I of 
+          None \<Rightarrow> False 
+        | Some s' \<Rightarrow> s' \<in> G)"
+      
+    lemma check_plan_correct: "check_plan \<pi>s \<longleftrightarrow> valid_plan \<pi>s"
+      unfolding check_plan_def valid_plan_def 
+      by (auto split: option.split simp: simulate_plan_correct[symmetric])
+      
+  end  
+    
+  fun simulate_plan' :: "ast_problem \<Rightarrow> plan \<Rightarrow> state \<rightharpoonup> state" where
+    "simulate_plan' problem [] s = Some s"
+  | "simulate_plan' problem (\<pi>#\<pi>s) s = (
+      if enabled' problem \<pi> s then
+        let s = execute' problem \<pi> s in
+        simulate_plan' problem \<pi>s s
+      else
+        None
+    )"  
+    
+  text \<open>Avoiding duplicate lookup.\<close>
+  (*[code]  *)  
+  lemma simulate_plan'_code[code]:
+    "simulate_plan' problem [] s = Some s"
+    "simulate_plan' problem (\<pi>#\<pi>s) s = (
+      case lookup_operator' problem \<pi> of
+        None \<Rightarrow> None
+      | Some \<pi> \<Rightarrow> 
+          if enabled_opr' \<pi> s then 
+            simulate_plan' problem \<pi>s (execute_opr' \<pi> s)
+          else None
+    )"
+    by (auto simp: enabled'_def execute'_def split: option.split)
+    
+    
+  definition initial_state' :: "ast_problem \<Rightarrow> state" where
+    "initial_state' problem \<equiv> let astI = ast_problem.astI problem in (
+       \<lambda>v. if v<length astI then Some (astI!v) else None
+     )"
+      
+  definition check_plan' :: "ast_problem \<Rightarrow> plan \<Rightarrow> bool" where
+    "check_plan' problem \<pi>s = (
+      case simulate_plan' problem \<pi>s (initial_state' problem) of 
+        None \<Rightarrow> False 
+      | Some s' \<Rightarrow> match_pres (ast_problem.astG problem) s')"
+      
+      
+  context wf_ast_problem 
+  begin  
+    
+    lemma simulate_plan'_correct:
+      assumes "s\<in>valid_states"
+      shows "simulate_plan' problem \<pi>s s = simulate_plan \<pi>s s"
+      using assms
+      apply (induction \<pi>s s rule: simulate_plan.induct)
+      apply (auto simp: enabled'_correct execute'_correct execute_preserves_valid)  
+      done
+        
+    lemma simulate_plan'_correct_paper: (* For presentation in paper. 
+        Summarizing intermediate refinement step. *)
+      assumes "s\<in>valid_states"
+      shows "simulate_plan' problem \<pi>s s = Some s'
+            \<longleftrightarrow> path_to s \<pi>s s'"
+      using simulate_plan'_correct[OF assms] simulate_plan_correct by simp
+      
+        
+    lemma initial_state'_correct: 
+      "initial_state' problem = I"
+      unfolding initial_state'_def I_def by (auto simp: Let_def)
+
+    lemma check_plan'_correct:
+      "check_plan' problem \<pi>s = check_plan \<pi>s"
+    proof -
+      have D: "distinct (map fst astG)" using wf_goal unfolding wf_partial_state_def by auto 
+        
+      have S'V: "s'\<in>valid_states" if "simulate_plan \<pi>s I = Some s'" for s'
+        using that by (auto simp: simulate_plan_correct path_to_pres_valid[OF I_valid])
+          
+      show ?thesis
+        unfolding check_plan'_def check_plan_def
+        by (auto 
+            split: option.splits 
+            simp: initial_state'_correct simulate_plan'_correct[OF I_valid]
+            simp: match_pres_correct[OF D S'V] G_def
+            )
+    qed  
+        
+  end
+
+    
+  (* Overall checker *)  
+    
+  definition verify_plan :: "ast_problem \<Rightarrow> plan \<Rightarrow> String.literal + unit" where
+    "verify_plan problem \<pi>s = (
+      if ast_problem.well_formed problem then
+        if check_plan' problem \<pi>s then Inr () else Inl (STR ''Invalid plan'')
+      else Inl (STR ''Problem not well formed'')
+    )"
+
+  lemma verify_plan_correct:
+    "verify_plan problem \<pi>s = Inr () 
+    \<longleftrightarrow> ast_problem.well_formed problem \<and> ast_problem.valid_plan problem \<pi>s"
+  proof -
+    {
+      assume "ast_problem.well_formed problem"
+      then interpret wf_ast_problem by unfold_locales
+          
+      from check_plan'_correct check_plan_correct 
+      have "check_plan' problem \<pi>s = valid_plan \<pi>s" by simp
+    } 
+    then show ?thesis  
+      unfolding verify_plan_def
+      by auto  
+  qed
+
+  definition nat_opt_of_integer :: "integer \<Rightarrow> nat option" where
+       "nat_opt_of_integer i = (if (i \<ge> 0) then Some (nat_of_integer i) else None)"
+
+  (*Export functions, which includes constructors*)
+  export_code verify_plan nat_of_integer integer_of_nat nat_opt_of_integer Inl Inr String.explode String.implode
+    in SML
+    module_name SASP_Checker_Exported
+    file "code/SASP_Checker_Exported.sml"
+
+end

formal/afp/AI_Planning_Languages_Semantics/SASP_Semantics.thy

@@ -0,0 +1,228 @@
+theory SASP_Semantics
+imports Main
+begin
+
+section \<open>Semantics of Fast-Downward's Multi-Valued Planning Tasks Language\<close>
+
+subsection \<open>Syntax\<close>
+  type_synonym name = string
+  type_synonym ast_variable = "name \<times> nat option \<times> name list" (* var name, axiom layer, atom names *)
+  type_synonym ast_variable_section = "ast_variable list"
+  type_synonym ast_initial_state = "nat list"
+  type_synonym ast_goal = "(nat \<times> nat) list"
+  type_synonym ast_precond = "(nat \<times> nat)"
+  type_synonym ast_effect = "ast_precond list \<times> nat \<times> nat option \<times> nat"
+  type_synonym ast_operator = "name \<times> ast_precond list \<times> ast_effect list \<times> nat"
+  type_synonym ast_operator_section = "ast_operator list"
+  
+  type_synonym ast_problem = 
+    "ast_variable_section \<times> ast_initial_state \<times> ast_goal \<times> ast_operator_section"
+
+  type_synonym plan = "name list"
+    
+  subsubsection \<open>Well-Formedness\<close>
+    
+  locale ast_problem =
+    fixes problem :: ast_problem
+  begin    
+    definition astDom :: ast_variable_section (* TODO: Dom \<rightarrow> Vars, D \<rightarrow> X*)
+      where "astDom \<equiv> case problem of (D,I,G,\<delta>) \<Rightarrow> D"
+    definition astI :: ast_initial_state
+      where "astI \<equiv> case problem of (D,I,G,\<delta>) \<Rightarrow> I"
+    definition astG :: ast_goal
+      where "astG \<equiv> case problem of (D,I,G,\<delta>) \<Rightarrow> G"
+    definition ast\<delta> :: ast_operator_section
+      where "ast\<delta> \<equiv> case problem of (D,I,G,\<delta>) \<Rightarrow> \<delta>"
+    
+    definition "numVars \<equiv> length astDom"
+    definition "numVals x \<equiv> length (snd (snd (astDom!x)))"
+
+    definition "wf_partial_state ps \<equiv> 
+        distinct (map fst ps) 
+      \<and> (\<forall>(x,v) \<in> set ps. x < numVars \<and> v < numVals x)"
+      
+    definition wf_operator :: "ast_operator \<Rightarrow> bool" 
+      where "wf_operator \<equiv> \<lambda>(name, pres, effs, cost). 
+        wf_partial_state pres 
+      \<and> distinct (map (\<lambda>(_, v, _, _). v) effs) \<comment> \<open>This may be too restrictive\<close>
+      \<and> (\<forall>(epres,x,vp,v)\<in>set effs. 
+          wf_partial_state epres 
+        \<and> x < numVars \<and> v < numVals x  
+        \<and> (case vp of None \<Rightarrow> True | Some v \<Rightarrow> v<numVals x)
+        )
+    "
+      
+    definition "well_formed \<equiv> 
+      \<comment> \<open>Initial state\<close>
+      length astI = numVars
+    \<and> (\<forall>x<numVars. astI!x < numVals x)
+
+      \<comment> \<open>Goal\<close>
+    \<and> wf_partial_state astG
+
+    \<comment> \<open>Operators\<close>
+    \<and> (distinct (map fst ast\<delta>))
+    \<and> (\<forall>\<pi>\<in>set ast\<delta>. wf_operator \<pi>)
+    "
+      
+  end  
+  
+  locale wf_ast_problem = ast_problem +
+    assumes wf: well_formed
+  begin
+    lemma wf_initial: 
+      "length astI = numVars" 
+      "\<forall>x<numVars. astI!x < numVals x"
+      using wf unfolding well_formed_def by auto
+
+    lemma wf_goal: "wf_partial_state astG"    
+      using wf unfolding well_formed_def by auto
+
+    lemma wf_operators: 
+      "distinct (map fst ast\<delta>)"
+      "\<forall>\<pi>\<in>set ast\<delta>. wf_operator \<pi>"
+      using wf unfolding well_formed_def by auto
+  end      
+    
+    
+  subsection \<open>Semantics as Transition System\<close>  
+    
+  type_synonym state = "nat \<rightharpoonup> nat"
+  type_synonym pstate = "nat \<rightharpoonup> nat"
+    
+    
+  context ast_problem
+  begin    
+    
+    definition Dom :: "nat set" where "Dom = {0..<numVars}"
+
+    definition range_of_var where "range_of_var x \<equiv> {0..<numVals x}"
+
+    definition valid_states :: "state set" where "valid_states \<equiv> {
+      s. dom s = Dom \<and> (\<forall>x\<in>Dom. the (s x) \<in> range_of_var x)
+    }"
+
+    definition I :: state 
+      where "I v \<equiv> if v<length astI then Some (astI!v) else None"
+
+    definition subsuming_states :: "pstate \<Rightarrow> state set"
+      where "subsuming_states partial \<equiv> { s\<in>valid_states. partial \<subseteq>\<^sub>m s }"
+
+    definition G :: "state set" 
+      where "G \<equiv> subsuming_states (map_of astG)"
+end
+
+    definition implicit_pres :: "ast_effect list \<Rightarrow> ast_precond list" where 
+      "implicit_pres effs \<equiv> 
+      map (\<lambda>(_,v,vpre,_). (v,the vpre))
+          (filter (\<lambda>(_,_,vpre,_). vpre\<noteq>None) effs)"
+
+context ast_problem
+begin
+
+    definition lookup_operator :: "name \<Rightarrow> ast_operator option" where
+      "lookup_operator name \<equiv> find (\<lambda>(n,_,_,_). n=name) ast\<delta>"
+
+    definition enabled :: "name \<Rightarrow> state \<Rightarrow> bool"
+      where "enabled name s \<equiv>
+        case lookup_operator name of
+          Some (_,pres,effs,_) \<Rightarrow> 
+              s\<in>subsuming_states (map_of pres)
+            \<and> s\<in>subsuming_states (map_of (implicit_pres effs))
+        | None \<Rightarrow> False"
+      
+    definition eff_enabled :: "state \<Rightarrow> ast_effect \<Rightarrow> bool" where
+      "eff_enabled s \<equiv> \<lambda>(pres,_,_,_). s\<in>subsuming_states (map_of pres)"
+
+    definition execute :: "name \<Rightarrow> state \<Rightarrow> state" where
+      "execute name s \<equiv> 
+        case lookup_operator name of
+          Some (_,_,effs,_) \<Rightarrow>
+            s ++ map_of (map (\<lambda>(_,x,_,v). (x,v)) (filter (eff_enabled s) effs))
+        | None \<Rightarrow> undefined                                    
+        "
+
+    fun path_to where
+      "path_to s [] s' \<longleftrightarrow> s'=s"
+    | "path_to s (\<pi>#\<pi>s) s' \<longleftrightarrow> enabled \<pi> s \<and> path_to (execute \<pi> s) \<pi>s s'"
+
+    definition valid_plan :: "plan \<Rightarrow> bool" 
+      where "valid_plan \<pi>s \<equiv> \<exists>s'\<in>G. path_to I \<pi>s s'"
+
+
+  end
+
+  (*
+    Next steps:
+      * well-formed stuff
+      * Executable SAS+ validator (well_formed and execute function)
+
+  *)  
+    
+  subsubsection \<open>Preservation of well-formedness\<close>  
+  context wf_ast_problem 
+  begin      
+    lemma I_valid: "I \<in> valid_states"
+      using wf_initial 
+      unfolding valid_states_def Dom_def I_def range_of_var_def
+      by (auto split:if_splits)
+      
+    lemma lookup_operator_wf:
+      assumes "lookup_operator name = Some \<pi>"
+      shows "wf_operator \<pi>" "fst \<pi> = name"  
+    proof -
+      obtain name' pres effs cost where [simp]: "\<pi>=(name',pres,effs,cost)" by (cases \<pi>)
+      hence [simp]: "name'=name" and IN_AST: "(name,pres,effs,cost) \<in> set ast\<delta>"
+        using assms
+        unfolding lookup_operator_def
+        apply -
+        apply (metis (mono_tags, lifting) case_prodD find_Some_iff)  
+        by (metis (mono_tags, lifting) case_prodD find_Some_iff nth_mem)  
+      
+      from IN_AST show WF: "wf_operator \<pi>" "fst \<pi> = name"   
+        unfolding enabled_def using wf_operators by auto
+    qed
+        
+        
+    lemma execute_preserves_valid:
+      assumes "s\<in>valid_states"  
+      assumes "enabled name s"  
+      shows "execute name s \<in> valid_states"  
+    proof -
+      from \<open>enabled name s\<close> obtain name' pres effs cost where
+        [simp]: "lookup_operator name = Some (name',pres,effs,cost)"
+        unfolding enabled_def by (auto split: option.splits)
+      from lookup_operator_wf[OF this] have WF: "wf_operator (name,pres,effs,cost)" by simp   
+      
+      have X1: "s ++ m \<in> valid_states" if "\<forall>x v. m x = Some v \<longrightarrow> x<numVars \<and> v<numVals x" for m
+        using that \<open>s\<in>valid_states\<close>
+        by (auto 
+            simp: valid_states_def Dom_def range_of_var_def map_add_def dom_def 
+            split: option.splits)  
+      
+      have X2: "x<numVars" "v<numVals x" 
+        if "map_of (map (\<lambda>(_, x, _, y). (x, y)) (filter (eff_enabled s) effs)) x = Some v"    
+        for x v
+      proof -
+        from that obtain epres vp where "(epres,x,vp,v) \<in> set effs"
+          by (auto dest!: map_of_SomeD)
+        with WF show "x<numVars" "v<numVals x"
+          unfolding wf_operator_def by auto
+      qed
+          
+      show ?thesis
+        using assms  
+        unfolding enabled_def execute_def 
+        by (auto intro!: X1 X2)
+    qed      
+    
+    lemma path_to_pres_valid:
+      assumes "s\<in>valid_states"
+      assumes "path_to s \<pi>s s'"
+      shows "s'\<in>valid_states"
+      using assms
+      by (induction s \<pi>s s' rule: path_to.induct) (auto simp: execute_preserves_valid)  
+      
+  end
+
+end

formal/afp/AI_Planning_Languages_Semantics/document/root.tex

@@ -0,0 +1,72 @@
+\documentclass[11pt,a4paper]{article}
+\usepackage[T1]{fontenc}
+\usepackage{isabelle,isabellesym}
+
+% further packages required for unusual symbols (see also
+% isabellesym.sty), use only when needed
+
+%\usepackage{amssymb}
+  %for \<leadsto>, \<box>, \<diamond>, \<sqsupset>, \<mho>, \<Join>,
+  %\<lhd>, \<lesssim>, \<greatersim>, \<lessapprox>, \<greaterapprox>,
+  %\<triangleq>, \<yen>, \<lozenge>
+
+%\usepackage{eurosym}
+  %for \<euro>
+
+%\usepackage[only,bigsqcap]{stmaryrd}
+  %for \<Sqinter>
+
+%\usepackage{eufrak}
+  %for \<AA> ... \<ZZ>, \<aa> ... \<zz> (also included in amssymb)
+
+%\usepackage{textcomp}
+  %for \<onequarter>, \<onehalf>, \<threequarters>, \<degree>, \<cent>,
+  %\<currency>
+
+\usepackage{wasysym}  
+  
+% this should be the last package used
+\usepackage{pdfsetup}
+
+% urls in roman style, theory text in math-similar italics
+\urlstyle{rm}
+\isabellestyle{it}
+
+% for uniform font size
+%\renewcommand{\isastyle}{\isastyleminor}
+
+
+\begin{document}
+
+\title{Semantics of AI Planning Languages}
+\author{Mohammad Abdulaziz and Peter Lammich\footnote{Author names are alphabetically ordered.}}
+
+% \subtitle{Proof Document}
+% \author{M. Abdulaziz \and P. Lammich}
+\date{}
+
+\maketitle
+
+This is an Isabelle/HOL formalisation of the semantics of the multi-valued planning tasks language that is used by the planning system Fast-Downward~\cite{helmert2006fast}, the STRIPS~\cite{fikes1971strips} fragment of the Planning Domain Definition Language~\cite{PDDLref} (PDDL), and the STRIPS soundness meta-theory developed by Lifschitz~\cite{lifschitz1987semantics}.
+It also contains formally verified checkers for checking the well-formedness of problems specified in either language as well the correctness of potential solutions.
+The formalisation in this entry was described in an earlier publication~\cite{ictai2018}.
+
+\tableofcontents
+
+\clearpage
+
+% sane default for proof documents
+\parindent 0pt\parskip 0.5ex
+
+% generated text of all theories
+\input{session}
+
+\bibliographystyle{abbrv}
+\bibliography{root}
+
+\end{document}
+
+%%% Local Variables:
+%%% mode: latex
+%%% TeX-master: t
+%%% End:

formal/afp/AODV/All.thy

@@ -0,0 +1,16 @@
+(*  Title:       All.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+    Author:      Peter Höfner, NICTA
+*)
+
+theory %invisible All
+imports Aodv_Loop_Freedom
+  "variants/a_norreqid/A_Aodv_Loop_Freedom"
+  "variants/b_fwdrreps/B_Aodv_Loop_Freedom"
+  "variants/c_gtobcast/C_Aodv_Loop_Freedom"
+  "variants/d_fwdrreqs/D_Aodv_Loop_Freedom"
+  "variants/e_all_abcd/E_Aodv_Loop_Freedom"
+begin
+
+end %invisible

formal/afp/AODV/Aodv.thy

@@ -0,0 +1,535 @@
+(*  Title:       Aodv.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "The AODV protocol"
+
+theory Aodv
+imports Aodv_Data Aodv_Message
+        AWN.AWN_SOS_Labels AWN.AWN_Invariants
+begin
+
+subsection "Data state"
+
+record state =
+  ip    :: "ip"
+  sn    :: "sqn"
+  rt    :: "rt" 
+  rreqs :: "(ip \<times> rreqid) set"
+  store :: "store"
+  (* all locals *)
+  msg    :: "msg"
+  data   :: "data"
+  dests  :: "ip \<rightharpoonup> sqn"
+  pre    :: "ip set"
+  rreqid :: "rreqid"
+  dip    :: "ip"
+  oip    :: "ip"
+  hops   :: "nat"
+  dsn    :: "sqn"
+  dsk    :: "k"
+  osn    :: "sqn"
+  sip    :: "ip"
+
+abbreviation aodv_init :: "ip \<Rightarrow> state"
+where "aodv_init i \<equiv> \<lparr>
+         ip = i,
+         sn = 1,
+         rt = Map.empty,
+         rreqs = {},
+         store = Map.empty,
+
+         msg    = (SOME x. True),
+         data   = (SOME x. True),
+         dests  = (SOME x. True),
+         pre    = (SOME x. True),
+         rreqid = (SOME x. True),
+         dip    = (SOME x. True),
+         oip    = (SOME x. True),
+         hops   = (SOME x. True),
+         dsn    = (SOME x. True),
+         dsk    = (SOME x. True),
+         osn    = (SOME x. True),
+         sip    = (SOME x. x \<noteq> i)
+       \<rparr>"
+
+lemma some_neq_not_eq [simp]: "\<not>((SOME x :: nat. x \<noteq> i) = i)"
+  by (subst some_eq_ex) (metis zero_neq_numeral)
+
+definition clear_locals :: "state \<Rightarrow> state"
+where "clear_locals \<xi> = \<xi> \<lparr>
+    msg    := (SOME x. True),
+    data   := (SOME x. True),
+    dests  := (SOME x. True),
+    pre    := (SOME x. True),
+    rreqid := (SOME x. True),
+    dip    := (SOME x. True),
+    oip    := (SOME x. True),
+    hops   := (SOME x. True),
+    dsn    := (SOME x. True),
+    dsk    := (SOME x. True),
+    osn    := (SOME x. True),
+    sip    := (SOME x. x \<noteq> ip \<xi>)
+  \<rparr>"
+
+lemma clear_locals_sip_not_ip [simp]: "\<not>(sip (clear_locals \<xi>) = ip \<xi>)"
+  unfolding clear_locals_def by simp
+
+lemma clear_locals_but_not_globals [simp]:
+  "ip (clear_locals \<xi>) = ip \<xi>"
+  "sn (clear_locals \<xi>) = sn \<xi>"
+  "rt (clear_locals \<xi>) = rt \<xi>"
+  "rreqs (clear_locals \<xi>) = rreqs \<xi>"
+  "store (clear_locals \<xi>) = store \<xi>"
+  unfolding clear_locals_def by auto
+
+subsection "Auxilliary message handling definitions"
+
+definition is_newpkt
+where "is_newpkt \<xi> \<equiv> case msg \<xi> of
+                       Newpkt data' dip' \<Rightarrow> { \<xi>\<lparr>data := data', dip := dip'\<rparr> }
+                     | _ \<Rightarrow> {}"
+
+definition is_pkt
+where "is_pkt \<xi> \<equiv> case msg \<xi> of
+                    Pkt data' dip' oip' \<Rightarrow> { \<xi>\<lparr> data := data', dip := dip', oip := oip' \<rparr> }
+                  | _ \<Rightarrow> {}"
+
+definition is_rreq
+where "is_rreq \<xi> \<equiv> case msg \<xi> of
+                     Rreq hops' rreqid' dip' dsn' dsk' oip' osn' sip' \<Rightarrow>
+                       { \<xi>\<lparr> hops := hops', rreqid := rreqid', dip := dip', dsn := dsn',
+                            dsk := dsk', oip := oip', osn := osn', sip := sip' \<rparr> }
+                   | _ \<Rightarrow> {}"
+
+lemma is_rreq_asm [dest!]:
+  assumes "\<xi>' \<in> is_rreq \<xi>"
+    shows "(\<exists>hops' rreqid' dip' dsn' dsk' oip' osn' sip'.
+               msg \<xi> = Rreq hops' rreqid' dip' dsn' dsk' oip' osn' sip' \<and>
+               \<xi>' = \<xi>\<lparr> hops := hops', rreqid := rreqid', dip := dip', dsn := dsn',
+                       dsk := dsk', oip := oip', osn := osn', sip := sip' \<rparr>)"
+  using assms unfolding is_rreq_def
+  by (cases "msg \<xi>") simp_all
+
+definition is_rrep
+where "is_rrep \<xi> \<equiv> case msg \<xi> of
+                     Rrep hops' dip' dsn' oip' sip' \<Rightarrow>
+                       { \<xi>\<lparr> hops := hops', dip := dip', dsn := dsn', oip := oip', sip := sip' \<rparr> }
+                   | _ \<Rightarrow> {}"
+
+lemma is_rrep_asm [dest!]:
+  assumes "\<xi>' \<in> is_rrep \<xi>"
+    shows "(\<exists>hops' dip' dsn' oip' sip'.
+               msg \<xi> = Rrep hops' dip' dsn' oip' sip' \<and>
+               \<xi>' = \<xi>\<lparr> hops := hops', dip := dip', dsn := dsn', oip := oip', sip := sip' \<rparr>)"
+  using assms unfolding is_rrep_def
+  by (cases "msg \<xi>") simp_all
+
+definition is_rerr
+where "is_rerr \<xi> \<equiv> case msg \<xi> of
+                     Rerr dests' sip' \<Rightarrow> { \<xi>\<lparr> dests := dests', sip := sip' \<rparr> }
+                   | _ \<Rightarrow> {}"
+
+lemma is_rerr_asm [dest!]:
+  assumes "\<xi>' \<in> is_rerr \<xi>"
+    shows "(\<exists>dests' sip'.
+               msg \<xi> = Rerr dests' sip' \<and>
+               \<xi>' = \<xi>\<lparr> dests := dests', sip := sip' \<rparr>)"
+  using assms unfolding is_rerr_def
+  by (cases "msg \<xi>") simp_all
+
+lemmas is_msg_defs =
+  is_rerr_def is_rrep_def is_rreq_def is_pkt_def is_newpkt_def
+
+lemma is_msg_inv_ip [simp]:
+  "\<xi>' \<in> is_rerr \<xi>   \<Longrightarrow> ip \<xi>' = ip \<xi>"
+  "\<xi>' \<in> is_rrep \<xi>   \<Longrightarrow> ip \<xi>' = ip \<xi>"
+  "\<xi>' \<in> is_rreq \<xi>   \<Longrightarrow> ip \<xi>' = ip \<xi>"
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> ip \<xi>' = ip \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> ip \<xi>' = ip \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+lemma is_msg_inv_sn [simp]:
+  "\<xi>' \<in> is_rerr \<xi>   \<Longrightarrow> sn \<xi>' = sn \<xi>"
+  "\<xi>' \<in> is_rrep \<xi>   \<Longrightarrow> sn \<xi>' = sn \<xi>"
+  "\<xi>' \<in> is_rreq \<xi>   \<Longrightarrow> sn \<xi>' = sn \<xi>"
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> sn \<xi>' = sn \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> sn \<xi>' = sn \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+lemma is_msg_inv_rt [simp]:
+  "\<xi>' \<in> is_rerr \<xi>   \<Longrightarrow> rt \<xi>' = rt \<xi>"
+  "\<xi>' \<in> is_rrep \<xi>   \<Longrightarrow> rt \<xi>' = rt \<xi>"
+  "\<xi>' \<in> is_rreq \<xi>   \<Longrightarrow> rt \<xi>' = rt \<xi>"
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> rt \<xi>' = rt \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> rt \<xi>' = rt \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+lemma is_msg_inv_rreqs [simp]:
+  "\<xi>' \<in> is_rerr \<xi>   \<Longrightarrow> rreqs \<xi>' = rreqs \<xi>"
+  "\<xi>' \<in> is_rrep \<xi>   \<Longrightarrow> rreqs \<xi>' = rreqs \<xi>"
+  "\<xi>' \<in> is_rreq \<xi>   \<Longrightarrow> rreqs \<xi>' = rreqs \<xi>"
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> rreqs \<xi>' = rreqs \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> rreqs \<xi>' = rreqs \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+lemma is_msg_inv_store [simp]:
+  "\<xi>' \<in> is_rerr \<xi>   \<Longrightarrow> store \<xi>' = store \<xi>"
+  "\<xi>' \<in> is_rrep \<xi>   \<Longrightarrow> store \<xi>' = store \<xi>"
+  "\<xi>' \<in> is_rreq \<xi>   \<Longrightarrow> store \<xi>' = store \<xi>"
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> store \<xi>' = store \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> store \<xi>' = store \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+lemma is_msg_inv_sip [simp]:
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> sip \<xi>' = sip \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> sip \<xi>' = sip \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+subsection "The protocol process"
+
+datatype pseqp =
+    PAodv
+  | PNewPkt
+  | PPkt
+  | PRreq
+  | PRrep
+  | PRerr
+
+fun nat_of_seqp :: "pseqp \<Rightarrow> nat"
+where
+  "nat_of_seqp PAodv  = 1"
+| "nat_of_seqp PPkt   = 2"
+| "nat_of_seqp PNewPkt   = 3"
+| "nat_of_seqp PRreq  = 4"
+| "nat_of_seqp PRrep  = 5"
+| "nat_of_seqp PRerr  = 6"
+
+instantiation "pseqp" :: ord
+begin
+definition less_eq_seqp [iff]: "l1 \<le> l2 = (nat_of_seqp l1 \<le> nat_of_seqp l2)"
+definition less_seqp [iff]:    "l1 < l2 = (nat_of_seqp l1 < nat_of_seqp l2)"
+instance ..
+end
+
+abbreviation AODV
+where
+  "AODV \<equiv> \<lambda>_. \<lbrakk>clear_locals\<rbrakk> call(PAodv)"
+
+abbreviation PKT
+where
+  "PKT args \<equiv>
+
+     \<lbrakk>\<xi>. let (data, dip, oip) = args \<xi> in
+         (clear_locals \<xi>) \<lparr> data := data, dip := dip, oip := oip \<rparr>\<rbrakk>
+     call(PPkt)"
+abbreviation NEWPKT
+where
+  "NEWPKT args \<equiv>
+     \<lbrakk>\<xi>. let (data, dip) = args \<xi> in
+         (clear_locals \<xi>) \<lparr> data := data, dip := dip \<rparr>\<rbrakk>
+     call(PNewPkt)"
+
+abbreviation RREQ
+where
+  "RREQ args \<equiv>
+     \<lbrakk>\<xi>. let (hops, rreqid, dip, dsn, dsk, oip, osn, sip) = args \<xi> in
+         (clear_locals \<xi>) \<lparr> hops := hops, rreqid := rreqid, dip := dip,
+                            dsn := dsn, dsk := dsk, oip := oip,
+                            osn := osn, sip := sip \<rparr>\<rbrakk>
+     call(PRreq)"
+
+abbreviation RREP
+where
+  "RREP args \<equiv>
+     \<lbrakk>\<xi>. let (hops, dip, dsn, oip, sip) = args \<xi> in
+         (clear_locals \<xi>) \<lparr> hops := hops, dip := dip, dsn := dsn,
+                            oip := oip, sip := sip \<rparr>\<rbrakk>
+     call(PRrep)"
+
+abbreviation RERR
+where
+  "RERR args \<equiv>
+     \<lbrakk>\<xi>. let (dests, sip) = args \<xi> in
+         (clear_locals \<xi>) \<lparr> dests := dests, sip := sip \<rparr>\<rbrakk>
+     call(PRerr)"
+
+fun \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V :: "(state, msg, pseqp, pseqp label) seqp_env"
+where
+  "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PAodv = labelled PAodv (
+     receive(\<lambda>msg' \<xi>. \<xi> \<lparr> msg := msg' \<rparr>).
+     (    \<langle>is_newpkt\<rangle> NEWPKT(\<lambda>\<xi>. (data \<xi>, ip \<xi>))
+       \<oplus> \<langle>is_pkt\<rangle> PKT(\<lambda>\<xi>. (data \<xi>, dip \<xi>, oip \<xi>))
+       \<oplus> \<langle>is_rreq\<rangle>
+            \<lbrakk>\<xi>. \<xi> \<lparr>rt := update (rt \<xi>) (sip \<xi>) (0, unk, val, 1, sip \<xi>, {}) \<rparr>\<rbrakk>
+            RREQ(\<lambda>\<xi>. (hops \<xi>, rreqid \<xi>, dip \<xi>, dsn \<xi>, dsk \<xi>, oip \<xi>, osn \<xi>, sip \<xi>))
+       \<oplus> \<langle>is_rrep\<rangle>
+            \<lbrakk>\<xi>. \<xi> \<lparr>rt := update (rt \<xi>) (sip \<xi>) (0, unk, val, 1, sip \<xi>, {}) \<rparr>\<rbrakk>
+            RREP(\<lambda>\<xi>. (hops \<xi>, dip \<xi>, dsn \<xi>, oip \<xi>, sip \<xi>))
+       \<oplus> \<langle>is_rerr\<rangle>
+            \<lbrakk>\<xi>. \<xi> \<lparr>rt := update (rt \<xi>) (sip \<xi>) (0, unk, val, 1, sip \<xi>, {}) \<rparr>\<rbrakk>
+            RERR(\<lambda>\<xi>. (dests \<xi>, sip \<xi>))
+     )
+     \<oplus> \<langle>\<lambda>\<xi>. { \<xi>\<lparr> dip := dip \<rparr> | dip. dip \<in> qD(store \<xi>) \<inter> vD(rt \<xi>) }\<rangle>
+          \<lbrakk>\<xi>. \<xi> \<lparr> data := hd(\<sigma>\<^bsub>queue\<^esub>(store \<xi>, dip \<xi>)) \<rparr>\<rbrakk>
+          unicast(\<lambda>\<xi>. the (nhop (rt \<xi>) (dip \<xi>)), \<lambda>\<xi>. pkt(data \<xi>, dip \<xi>, ip \<xi>)).
+            \<lbrakk>\<xi>. \<xi> \<lparr> store := the (drop (dip \<xi>) (store \<xi>)) \<rparr>\<rbrakk>
+            AODV()
+          \<triangleright> \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if (rip \<in> vD (rt \<xi>) \<and> nhop (rt \<xi>) rip = nhop (rt \<xi>) (dip \<xi>))
+                                     then Some (inc (sqn (rt \<xi>) rip)) else None) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                                     then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+             groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)). AODV()
+     \<oplus> \<langle>\<lambda>\<xi>. { \<xi>\<lparr> dip := dip \<rparr>
+             | dip. dip \<in> qD(store \<xi>) - vD(rt \<xi>) \<and> the (\<sigma>\<^bsub>p-flag\<^esub>(store \<xi>, dip)) = req }\<rangle>
+         \<lbrakk>\<xi>. \<xi> \<lparr> store := unsetRRF (store \<xi>) (dip \<xi>) \<rparr>\<rbrakk>
+         \<lbrakk>\<xi>. \<xi> \<lparr> sn := inc (sn \<xi>) \<rparr>\<rbrakk>
+         \<lbrakk>\<xi>. \<xi> \<lparr> rreqid := nrreqid (rreqs \<xi>) (ip \<xi>) \<rparr>\<rbrakk>
+         \<lbrakk>\<xi>. \<xi> \<lparr> rreqs := rreqs \<xi> \<union> {(ip \<xi>, rreqid \<xi>)} \<rparr>\<rbrakk>
+         broadcast(\<lambda>\<xi>. rreq(0, rreqid \<xi>, dip \<xi>, sqn (rt \<xi>) (dip \<xi>), sqnf (rt \<xi>) (dip \<xi>),
+                            ip \<xi>, sn \<xi>, ip \<xi>)). AODV())"
+
+|  "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PNewPkt = labelled PNewPkt (
+     \<langle>\<xi>. dip \<xi> = ip \<xi>\<rangle>
+        deliver(\<lambda>\<xi>. data \<xi>).AODV()
+     \<oplus> \<langle>\<xi>. dip \<xi> \<noteq> ip \<xi>\<rangle>
+        \<lbrakk>\<xi>. \<xi> \<lparr> store := add (data \<xi>) (dip \<xi>) (store \<xi>) \<rparr>\<rbrakk>
+        AODV())"
+
+| "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PPkt = labelled PPkt (
+     \<langle>\<xi>. dip \<xi> = ip \<xi>\<rangle>
+        deliver(\<lambda>\<xi>. data \<xi>).AODV()
+     \<oplus> \<langle>\<xi>. dip \<xi> \<noteq> ip \<xi>\<rangle>
+     (
+       \<langle>\<xi>. dip \<xi> \<in> vD (rt \<xi>)\<rangle>
+         unicast(\<lambda>\<xi>. the (nhop (rt \<xi>) (dip \<xi>)), \<lambda>\<xi>. pkt(data \<xi>, dip \<xi>, oip \<xi>)).AODV()
+         \<triangleright>
+           \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if (rip \<in> vD (rt \<xi>) \<and> nhop (rt \<xi>) rip = nhop (rt \<xi>) (dip \<xi>))
+                                   then Some (inc (sqn (rt \<xi>) rip)) else None) \<rparr>\<rbrakk>
+           \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+           \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+           \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+           \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                                   then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+           groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)).AODV()
+       \<oplus> \<langle>\<xi>. dip \<xi> \<notin> vD (rt \<xi>)\<rangle>
+       (
+           \<langle>\<xi>. dip \<xi> \<in> iD (rt \<xi>)\<rangle>
+             groupcast(\<lambda>\<xi>. the (precs (rt \<xi>) (dip \<xi>)),
+                       \<lambda>\<xi>. rerr([dip \<xi> \<mapsto> sqn (rt \<xi>) (dip \<xi>)], ip \<xi>)). AODV()
+           \<oplus> \<langle>\<xi>. dip \<xi> \<notin> iD (rt \<xi>)\<rangle>
+              AODV()
+       )
+     ))"
+
+| "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRreq = labelled PRreq (
+     \<langle>\<xi>. (oip \<xi>, rreqid \<xi>) \<in> rreqs \<xi>\<rangle>
+       AODV()
+     \<oplus> \<langle>\<xi>. (oip \<xi>, rreqid \<xi>) \<notin> rreqs \<xi>\<rangle>
+       \<lbrakk>\<xi>. \<xi> \<lparr> rt := update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, hops \<xi> + 1, sip \<xi>, {}) \<rparr>\<rbrakk>
+       \<lbrakk>\<xi>. \<xi> \<lparr> rreqs := rreqs \<xi> \<union> {(oip \<xi>, rreqid \<xi>)} \<rparr>\<rbrakk>
+       (
+         \<langle>\<xi>. dip \<xi> = ip \<xi>\<rangle>
+           \<lbrakk>\<xi>. \<xi> \<lparr> sn := max (sn \<xi>) (dsn \<xi>) \<rparr>\<rbrakk>
+           unicast(\<lambda>\<xi>. the (nhop (rt \<xi>) (oip \<xi>)), \<lambda>\<xi>. rrep(0, dip \<xi>, sn \<xi>, oip \<xi>, ip \<xi>)).AODV()
+           \<triangleright>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if (rip \<in> vD (rt \<xi>) \<and> nhop (rt \<xi>) rip = nhop (rt \<xi>) (oip \<xi>))
+                                     then Some (inc (sqn (rt \<xi>) rip)) else None) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                                     then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+             groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)).AODV()
+         \<oplus> \<langle>\<xi>. dip \<xi> \<noteq> ip \<xi>\<rangle>
+         (
+           \<langle>\<xi>. dip \<xi> \<in> vD (rt \<xi>) \<and> dsn \<xi> \<le> sqn (rt \<xi>) (dip \<xi>) \<and> sqnf (rt \<xi>) (dip \<xi>) = kno\<rangle>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := the (addpreRT (rt \<xi>) (dip \<xi>) {sip \<xi>}) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := the (addpreRT (rt \<xi>) (oip \<xi>) {the (nhop (rt \<xi>) (dip \<xi>))}) \<rparr>\<rbrakk> 
+             unicast(\<lambda>\<xi>. the (nhop (rt \<xi>) (oip \<xi>)), \<lambda>\<xi>. rrep(the (dhops (rt \<xi>) (dip \<xi>)), dip \<xi>,
+                         sqn (rt \<xi>) (dip \<xi>), oip \<xi>, ip \<xi>)).
+             AODV()
+           \<triangleright>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if (rip \<in> vD (rt \<xi>) \<and> nhop (rt \<xi>) rip = nhop (rt \<xi>) (oip \<xi>))
+                                     then Some (inc (sqn (rt \<xi>) rip)) else None) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                                     then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+             groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)).AODV()
+           \<oplus> \<langle>\<xi>. dip \<xi> \<notin> vD (rt \<xi>) \<or> sqn (rt \<xi>) (dip \<xi>) < dsn \<xi> \<or> sqnf (rt \<xi>) (dip \<xi>) = unk\<rangle>
+             broadcast(\<lambda>\<xi>. rreq(hops \<xi> + 1, rreqid \<xi>, dip \<xi>, max (sqn (rt \<xi>) (dip \<xi>)) (dsn \<xi>),
+                                dsk \<xi>, oip \<xi>, osn \<xi>, ip \<xi>)).
+             AODV()
+         )
+       ))"
+
+| "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRrep = labelled PRrep (
+     \<langle>\<xi>. rt \<xi> \<noteq> update (rt \<xi>) (dip \<xi>) (dsn \<xi>, kno, val, hops \<xi> + 1, sip \<xi>, {}) \<rangle>
+     (
+       \<lbrakk>\<xi>. \<xi> \<lparr> rt := update (rt \<xi>) (dip \<xi>) (dsn \<xi>, kno, val, hops \<xi> + 1, sip \<xi>, {}) \<rparr> \<rbrakk>
+       (
+         \<langle>\<xi>. oip \<xi> = ip \<xi> \<rangle>
+            AODV()
+         \<oplus> \<langle>\<xi>. oip \<xi> \<noteq> ip \<xi> \<rangle>
+         (
+           \<langle>\<xi>. oip \<xi> \<in> vD (rt \<xi>)\<rangle>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := the (addpreRT (rt \<xi>) (dip \<xi>) {the (nhop (rt \<xi>) (oip \<xi>))}) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := the (addpreRT (rt \<xi>) (the (nhop (rt \<xi>) (dip \<xi>)))
+                                               {the (nhop (rt \<xi>) (oip \<xi>))}) \<rparr>\<rbrakk> 
+             unicast(\<lambda>\<xi>. the (nhop (rt \<xi>) (oip \<xi>)), \<lambda>\<xi>. rrep(hops \<xi> + 1, dip \<xi>, dsn \<xi>, oip \<xi>, ip \<xi>)).
+             AODV()
+           \<triangleright>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if (rip \<in> vD (rt \<xi>) \<and> nhop (rt \<xi>) rip = nhop (rt \<xi>) (oip \<xi>))
+                                     then Some (inc (sqn (rt \<xi>) rip)) else None) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                                     then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+             groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)).AODV()
+           \<oplus> \<langle>\<xi>. oip \<xi> \<notin> vD (rt \<xi>)\<rangle>
+             AODV()
+         )
+       )
+     )
+     \<oplus> \<langle>\<xi>. rt \<xi> = update (rt \<xi>) (dip \<xi>) (dsn \<xi>, kno, val, hops \<xi> + 1, sip \<xi>, {}) \<rangle>
+         AODV()
+     )"
+
+| "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRerr = labelled PRerr (
+     \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. case (dests \<xi>) rip of None \<Rightarrow> None
+                       | Some rsn \<Rightarrow> if rip \<in> vD (rt \<xi>) \<and> the (nhop (rt \<xi>) rip) = sip \<xi>
+                                       \<and> sqn (rt \<xi>) rip < rsn then Some rsn else None) \<rparr>\<rbrakk>
+     \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+     \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+     \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+     \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                             then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+     groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)). AODV())"
+
+declare \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V.simps [simp del, code del]
+lemmas \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_simps [simp, code] = \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V.simps [simplified]
+
+fun \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton
+where
+    "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PAodv   = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PAodv)"
+  | "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PNewPkt = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PNewPkt)"
+  | "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PPkt    = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PPkt)"
+  | "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PRreq   = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRreq)"
+  | "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PRrep   = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRrep)"
+  | "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PRerr   = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRerr)"
+
+lemma \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton_wf [simp]:
+  "wellformed \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton"
+  proof (rule, intro allI)
+    fix pn pn'
+    show "call(pn') \<notin> stermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton pn)"
+      by (cases pn) simp_all
+  qed
+
+declare \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton.simps [simp del, code del]
+lemmas \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton_simps [simp, code]
+           = \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton.simps [simplified \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_simps seqp_skeleton.simps]
+
+lemma aodv_proc_cases [dest]:
+  fixes p pn
+  shows "p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pn) \<Longrightarrow>
+                                (p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PAodv) \<or> 
+                                 p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PNewPkt)  \<or>
+                                 p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PPkt)  \<or>
+                                 p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRreq) \<or>
+                                 p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRrep) \<or>
+                                 p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRerr))"
+  by (cases pn) simp_all
+
+definition \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V :: "ip \<Rightarrow> (state \<times> (state, msg, pseqp, pseqp label) seqp) set"
+where "\<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i \<equiv> {(aodv_init i, \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PAodv)}"
+
+abbreviation paodv
+  :: "ip \<Rightarrow> (state \<times> (state, msg, pseqp, pseqp label) seqp, msg seq_action) automaton"
+where
+  "paodv i \<equiv> \<lparr> init = \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i, trans = seqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V \<rparr>"
+
+lemma aodv_trans: "trans (paodv i) = seqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V"
+  by simp
+
+lemma aodv_control_within [simp]: "control_within \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (init (paodv i))"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by (rule control_withinI) (auto simp del: \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_simps)
+
+lemma aodv_wf [simp]:
+  "wellformed \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V"
+  proof (rule, intro allI)
+    fix pn pn'
+    show "call(pn') \<notin> stermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pn)"
+      by (cases pn) simp_all
+  qed
+
+lemmas aodv_labels_not_empty [simp] = labels_not_empty [OF aodv_wf]
+
+lemma aodv_ex_label [intro]: "\<exists>l. l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p"
+  by (metis aodv_labels_not_empty all_not_in_conv)
+
+lemma aodv_ex_labelE [elim]:
+  assumes "\<forall>l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p. P l p"
+      and "\<exists>p l. P l p \<Longrightarrow> Q"
+    shows "Q"
+  using assms by (metis aodv_ex_label)
+
+lemma aodv_simple_labels [simp]: "simple_labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V"
+  proof
+    fix pn p
+    assume "p\<in>subterms(\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pn)"
+    thus "\<exists>!l. labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p = {l}"
+    by (cases pn) (simp_all cong: seqp_congs | elim disjE)+
+  qed
+
+lemma \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_labels [simp]: "(\<xi>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i \<Longrightarrow>  labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p = {PAodv-:0}"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by simp
+
+lemma aodv_init_kD_empty [simp]:
+  "(\<xi>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i \<Longrightarrow> kD (rt \<xi>) = {}"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def kD_def by simp
+
+lemma aodv_init_sip_not_ip [simp]: "\<not>(sip (aodv_init i) = i)" by simp
+
+lemma aodv_init_sip_not_ip' [simp]:
+  assumes "(\<xi>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i"
+    shows "sip \<xi> \<noteq> ip \<xi>"
+  using assms unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by simp
+
+lemma aodv_init_sip_not_i [simp]:
+  assumes "(\<xi>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i"
+    shows "sip \<xi> \<noteq> i"
+  using assms unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by simp
+
+lemma clear_locals_sip_not_ip':
+  assumes "ip \<xi> = i"
+    shows "\<not>(sip (clear_locals \<xi>) = i)"
+  using assms by auto
+
+text \<open>Stop the simplifier from descending into process terms.\<close>
+declare seqp_congs [cong]
+
+text \<open>Configure the main invariant tactic for AODV.\<close>
+
+declare
+  \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_simps [cterms_env]
+  aodv_proc_cases [ctermsl_cases]
+  seq_invariant_ctermsI [OF aodv_wf aodv_control_within aodv_simple_labels aodv_trans,
+                            cterms_intros]
+  seq_step_invariant_ctermsI [OF aodv_wf aodv_control_within aodv_simple_labels aodv_trans,
+                                 cterms_intros]
+
+end

formal/afp/AODV/Aodv_Basic.thy

@@ -0,0 +1,44 @@
+(*  Title:       Aodv_Basic.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "Basic data types and constants"
+
+theory Aodv_Basic
+imports Main AWN.AWN_SOS
+begin
+
+text \<open>These definitions are shared with all variants.\<close>
+
+type_synonym rreqid = nat
+type_synonym sqn = nat
+
+datatype k = Known | Unknown
+abbreviation kno where "kno \<equiv> Known"
+abbreviation unk where "unk \<equiv> Unknown"
+
+datatype p = NoRequestRequired | RequestRequired
+abbreviation noreq where "noreq \<equiv> NoRequestRequired"
+abbreviation req where "req \<equiv> RequestRequired"
+
+datatype f = Valid | Invalid
+abbreviation val where "val \<equiv> Valid"
+abbreviation inv where "inv \<equiv> Invalid"
+
+lemma not_ks [simp]:                                      
+   "(x \<noteq> kno) = (x = unk)"
+   "(x \<noteq> unk) = (x = kno)"
+  by (cases x, clarsimp+)+
+
+lemma not_ps [simp]:
+  "(x \<noteq> noreq) = (x = req)"
+  "(x \<noteq> req) = (x = noreq)"
+  by (cases x, clarsimp+)+
+
+lemma not_ffs [simp]:
+  "(x \<noteq> val) = (x = inv)"
+  "(x \<noteq> inv) = (x = val)"
+  by (cases x, clarsimp+)+
+
+end

formal/afp/AODV/Aodv_Data.thy

@@ -0,0 +1,990 @@
+(*  Title:       Aodv_Data.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "Predicates and functions used in the AODV model"
+
+theory Aodv_Data
+imports Aodv_Basic
+begin
+
+subsection "Sequence Numbers"
+
+text \<open>Sequence numbers approximate the relative freshness of routing information.\<close>
+
+definition inc :: "sqn \<Rightarrow> sqn"
+  where "inc sn \<equiv> if sn = 0 then sn else sn + 1"
+
+lemma less_than_inc [simp]: "x \<le> inc x"
+  unfolding inc_def by simp
+
+lemma inc_minus_suc_0 [simp]:
+  "inc x - Suc 0 = x"
+  unfolding inc_def by simp
+
+lemma inc_never_one' [simp, intro]: "inc x \<noteq> Suc 0"
+  unfolding inc_def by simp
+
+lemma inc_never_one [simp, intro]: "inc x \<noteq> 1"
+  by simp
+
+subsection "Modelling Routes"
+
+text \<open>
+  A route is a 6-tuple, @{term "(dsn, dsk, flag, hops, nhip, pre)"} where
+  @{term dsn} is the `destination sequence number', @{term dsk} is the
+  `destination-sequence-number status', @{term flag} is the route status,
+  @{term hops} is the number of hops to the destination, @{term nhip} is the
+  next hop toward the destination, and @{term pre} is the set of `precursor nodes'--those
+  interested in hearing about changes to the route.
+\<close>
+
+type_synonym r = "sqn \<times> k \<times> f \<times> nat \<times> ip \<times> ip set"
+
+definition proj2 :: "r \<Rightarrow> sqn" ("\<pi>\<^sub>2")
+  where "\<pi>\<^sub>2 \<equiv> \<lambda>(dsn, _, _, _, _, _). dsn"
+
+definition proj3 :: "r \<Rightarrow> k" ("\<pi>\<^sub>3")
+  where "\<pi>\<^sub>3 \<equiv> \<lambda>(_, dsk, _, _, _, _). dsk"
+
+definition proj4 :: "r \<Rightarrow> f" ("\<pi>\<^sub>4")
+  where "\<pi>\<^sub>4 \<equiv> \<lambda>(_, _, flag, _, _, _). flag"
+
+definition proj5 :: "r \<Rightarrow> nat" ("\<pi>\<^sub>5")
+  where "\<pi>\<^sub>5 \<equiv> \<lambda>(_, _, _, hops, _, _). hops"
+
+definition proj6 :: "r \<Rightarrow> ip" ("\<pi>\<^sub>6")
+  where "\<pi>\<^sub>6 \<equiv> \<lambda>(_, _, _, _, nhip, _). nhip"
+
+definition proj7 :: "r \<Rightarrow> ip set" ("\<pi>\<^sub>7")
+  where "\<pi>\<^sub>7 \<equiv> \<lambda>(_, _, _, _, _, pre). pre"
+
+lemma projs [simp]:
+  "\<pi>\<^sub>2(dsn, dsk, flag, hops, nhip, pre) = dsn"
+  "\<pi>\<^sub>3(dsn, dsk, flag, hops, nhip, pre) = dsk"
+  "\<pi>\<^sub>4(dsn, dsk, flag, hops, nhip, pre) = flag"
+  "\<pi>\<^sub>5(dsn, dsk, flag, hops, nhip, pre) = hops"
+  "\<pi>\<^sub>6(dsn, dsk, flag, hops, nhip, pre) = nhip"
+  "\<pi>\<^sub>7(dsn, dsk, flag, hops, nhip, pre) = pre"
+  by (clarsimp simp: proj2_def proj3_def proj4_def
+                     proj5_def proj6_def proj7_def)+
+
+lemma proj3_pred [intro]: "\<lbrakk> P kno; P unk \<rbrakk> \<Longrightarrow> P (\<pi>\<^sub>3 x)"
+  by (rule k.induct)
+
+lemma proj4_pred [intro]: "\<lbrakk> P val; P inv \<rbrakk> \<Longrightarrow> P (\<pi>\<^sub>4 x)"
+  by (rule f.induct)
+
+lemma proj6_pair_snd [simp]:
+  fixes dsn' r
+  shows "\<pi>\<^sub>6 (dsn', snd (r)) = \<pi>\<^sub>6(r)"
+  by (cases r) simp
+
+subsection "Routing Tables"
+
+text \<open>Routing tables map ip addresses to route entries.\<close>
+
+type_synonym rt = "ip \<rightharpoonup> r"
+
+syntax
+  "_Sigma_route" :: "rt \<Rightarrow> ip \<rightharpoonup> r"  ("\<sigma>\<^bsub>route\<^esub>'(_, _')")
+
+translations
+ "\<sigma>\<^bsub>route\<^esub>(rt, dip)" => "rt dip" 
+
+definition sqn :: "rt \<Rightarrow> ip \<Rightarrow> sqn"
+  where "sqn rt dip \<equiv> case \<sigma>\<^bsub>route\<^esub>(rt, dip) of Some r \<Rightarrow> \<pi>\<^sub>2(r) | None \<Rightarrow> 0"
+
+definition sqnf :: "rt \<Rightarrow> ip \<Rightarrow> k"
+  where "sqnf rt dip \<equiv> case \<sigma>\<^bsub>route\<^esub>(rt, dip) of Some r \<Rightarrow> \<pi>\<^sub>3(r) | None \<Rightarrow> unk"
+
+abbreviation flag :: "rt \<Rightarrow> ip \<rightharpoonup> f"
+  where "flag rt dip \<equiv> map_option \<pi>\<^sub>4 (\<sigma>\<^bsub>route\<^esub>(rt, dip))"
+
+abbreviation dhops :: "rt \<Rightarrow> ip \<rightharpoonup> nat"
+   where "dhops rt dip \<equiv> map_option \<pi>\<^sub>5 (\<sigma>\<^bsub>route\<^esub>(rt, dip))"
+
+abbreviation nhop :: "rt \<Rightarrow> ip \<rightharpoonup> ip"
+   where "nhop rt dip \<equiv> map_option \<pi>\<^sub>6 (\<sigma>\<^bsub>route\<^esub>(rt, dip))"
+
+abbreviation precs :: "rt \<Rightarrow> ip \<rightharpoonup> ip set"
+   where "precs rt dip \<equiv> map_option \<pi>\<^sub>7 (\<sigma>\<^bsub>route\<^esub>(rt, dip))"
+
+definition vD :: "rt \<Rightarrow> ip set"
+  where "vD rt \<equiv> {dip. flag rt dip = Some val}"
+
+definition iD :: "rt \<Rightarrow> ip set"
+  where "iD rt \<equiv> {dip. flag rt dip = Some inv}"
+
+definition kD :: "rt \<Rightarrow> ip set"
+  where "kD rt \<equiv> {dip. rt dip \<noteq> None}"
+
+lemma kD_is_vD_and_iD: "kD rt = vD rt \<union> iD rt"
+  unfolding kD_def vD_def iD_def by auto
+
+lemma vD_iD_gives_kD [simp]:
+   "\<And>ip rt. ip \<in> vD rt \<Longrightarrow> ip \<in> kD rt"
+   "\<And>ip rt. ip \<in> iD rt \<Longrightarrow> ip \<in> kD rt"
+  unfolding kD_is_vD_and_iD by simp_all
+
+lemma kD_Some [dest]:
+    fixes dip rt
+  assumes "dip \<in> kD rt"
+    shows "\<exists>dsn dsk flag hops nhip pre.
+           \<sigma>\<^bsub>route\<^esub>(rt, dip) = Some (dsn, dsk, flag, hops, nhip, pre)"
+  using assms unfolding kD_def by simp
+
+lemma kD_None [dest]:
+    fixes dip rt
+  assumes "dip \<notin> kD rt"
+    shows "\<sigma>\<^bsub>route\<^esub>(rt, dip) = None"
+  using assms unfolding kD_def
+  by (metis (mono_tags) mem_Collect_eq)
+
+lemma vD_Some [dest]:
+    fixes dip rt
+  assumes "dip \<in> vD rt"
+    shows "\<exists>dsn dsk hops nhip pre.
+           \<sigma>\<^bsub>route\<^esub>(rt, dip) = Some (dsn, dsk, val, hops, nhip, pre)"
+  using assms unfolding vD_def by simp
+
+lemma vD_empty [simp]: "vD Map.empty = {}"
+  unfolding vD_def by simp
+
+lemma iD_Some [dest]:
+    fixes dip rt
+  assumes "dip \<in> iD rt"
+    shows "\<exists>dsn dsk hops nhip pre.
+           \<sigma>\<^bsub>route\<^esub>(rt, dip) = Some (dsn, dsk, inv, hops, nhip, pre)"
+  using assms unfolding iD_def by simp
+
+lemma val_is_vD [elim]:
+    fixes ip rt
+  assumes "ip\<in>kD(rt)"
+      and "the (flag rt ip) = val"
+    shows "ip\<in>vD(rt)"
+  using assms unfolding vD_def by auto
+
+lemma inv_is_iD [elim]:
+    fixes ip rt
+  assumes "ip\<in>kD(rt)"
+      and "the (flag rt ip) = inv"
+    shows "ip\<in>iD(rt)"
+  using assms unfolding iD_def by auto
+
+lemma iD_flag_is_inv [elim, simp]:
+    fixes ip rt
+  assumes "ip\<in>iD(rt)"
+    shows "the (flag rt ip) = inv"
+  proof -
+    from \<open>ip\<in>iD(rt)\<close> have "ip\<in>kD(rt)" by auto
+    with assms show ?thesis unfolding iD_def by auto
+  qed
+
+lemma kD_but_not_vD_is_iD [elim]:
+    fixes ip rt
+  assumes "ip\<in>kD(rt)"
+      and "ip\<notin>vD(rt)"
+    shows "ip\<in>iD(rt)"
+  proof -
+    from \<open>ip\<in>kD(rt)\<close> obtain dsn dsk f hops nhop pre
+      where rtip: "rt ip = Some (dsn, dsk, f, hops, nhop, pre)"
+       by (metis kD_Some)
+    from \<open>ip\<notin>vD(rt)\<close> have "f \<noteq> val"
+    proof (rule contrapos_nn)
+      assume "f = val"
+      with rtip have "the (flag rt ip) = val" by simp
+      with \<open>ip\<in>kD(rt)\<close> show "ip\<in>vD(rt)" ..
+    qed
+    with rtip have "the (flag rt ip)= inv" by simp  
+    with \<open>ip\<in>kD(rt)\<close> show "ip\<in>iD(rt)" ..
+  qed
+
+lemma vD_or_iD [elim]:
+    fixes ip rt
+  assumes "ip\<in>kD(rt)"
+      and "ip\<in>vD(rt) \<Longrightarrow> P rt ip"
+      and "ip\<in>iD(rt) \<Longrightarrow> P rt ip"
+    shows "P rt ip"
+  proof -
+    from \<open>ip\<in>kD(rt)\<close> have "ip\<in>vD(rt) \<union> iD(rt)"
+      by (simp add: kD_is_vD_and_iD)
+    thus ?thesis by (auto elim: assms(2-3))
+  qed
+
+lemma proj5_eq_dhops: "\<And>dip rt. dip\<in>kD(rt) \<Longrightarrow> \<pi>\<^sub>5(the (rt dip)) = the (dhops rt dip)"
+  unfolding sqn_def by (drule kD_Some) clarsimp
+
+lemma proj4_eq_flag: "\<And>dip rt. dip\<in>kD(rt) \<Longrightarrow> \<pi>\<^sub>4(the (rt dip)) = the (flag rt dip)"
+  unfolding sqn_def by (drule kD_Some) clarsimp
+
+lemma proj2_eq_sqn: "\<And>dip rt. dip\<in>kD(rt) \<Longrightarrow> \<pi>\<^sub>2(the (rt dip)) = sqn rt dip"
+  unfolding sqn_def by (drule kD_Some) clarsimp
+
+lemma kD_sqnf_is_proj3 [simp]:
+  "\<And>ip rt. ip\<in>kD(rt) \<Longrightarrow> sqnf rt ip = \<pi>\<^sub>3(the (rt ip))"
+  unfolding sqnf_def by auto
+
+lemma vD_flag_val [simp]:
+  "\<And>dip rt. dip \<in> vD (rt) \<Longrightarrow> the (flag rt dip) = val"
+  unfolding vD_def by clarsimp
+
+lemma kD_update [simp]:
+  "\<And>rt nip v. kD (rt(nip \<mapsto> v)) = insert nip (kD rt)"
+  unfolding kD_def by auto
+
+lemma kD_empty [simp]: "kD Map.empty = {}"
+  unfolding kD_def by simp
+
+lemma ip_equal_or_known [elim]:
+  fixes rt ip ip'
+  assumes "ip = ip' \<or> ip\<in>kD(rt)"
+      and "ip = ip' \<Longrightarrow> P rt ip ip'"
+      and "\<lbrakk> ip \<noteq> ip'; ip\<in>kD(rt)\<rbrakk> \<Longrightarrow> P rt ip ip'"
+    shows "P rt ip ip'"
+  using assms by auto
+
+subsection "Updating Routing Tables"
+
+text \<open>Routing table entries are modified through explicit functions.
+      The properties of these functions are important in invariant proofs.\<close>
+
+subsubsection "Updating Precursor Lists"
+
+definition addpre :: "r \<Rightarrow> ip set \<Rightarrow> r"
+  where "addpre r npre \<equiv> let (dsn, dsk, flag, hops, nhip, pre) = r in
+                          (dsn, dsk, flag, hops, nhip, pre \<union> npre)"
+
+lemma proj2_addpre:
+  fixes v pre
+  shows "\<pi>\<^sub>2(addpre v pre) = \<pi>\<^sub>2(v)"
+  unfolding addpre_def by (cases v) simp
+
+lemma proj3_addpre:
+  fixes v pre
+  shows "\<pi>\<^sub>3(addpre v pre) = \<pi>\<^sub>3(v)"
+  unfolding addpre_def by (cases v) simp
+
+lemma proj4_addpre:
+  fixes v pre
+  shows "\<pi>\<^sub>4(addpre v pre) = \<pi>\<^sub>4(v)"
+  unfolding addpre_def by (cases v) simp
+
+lemma proj5_addpre:
+  fixes v pre
+  shows "\<pi>\<^sub>5(addpre v pre) = \<pi>\<^sub>5(v)"
+  unfolding addpre_def by (cases v) simp
+
+lemma proj6_addpre:
+  fixes dsn dsk flag hops nhip pre npre
+  shows "\<pi>\<^sub>6(addpre v npre) = \<pi>\<^sub>6(v)"
+  unfolding addpre_def by (cases v) simp
+
+lemma proj7_addpre:
+  fixes dsn dsk flag hops nhip pre npre
+  shows "\<pi>\<^sub>7(addpre v npre) = \<pi>\<^sub>7(v) \<union> npre"
+  unfolding addpre_def by (cases v) simp
+
+lemma addpre_empty: "addpre r {} = r"
+  unfolding addpre_def by simp
+
+lemma addpre_r:
+  "addpre (dsn, dsk, fl, hops, nhip, pre) npre = (dsn, dsk, fl, hops, nhip, pre \<union> npre)"
+  unfolding addpre_def by simp
+
+lemmas addpre_simps [simp] = proj2_addpre proj3_addpre proj4_addpre proj5_addpre
+                             proj6_addpre proj7_addpre addpre_empty addpre_r
+
+definition addpreRT :: "rt \<Rightarrow> ip \<Rightarrow> ip set \<rightharpoonup> rt"
+  where "addpreRT rt dip npre \<equiv>
+           map_option (\<lambda>s. rt (dip \<mapsto> addpre s npre)) (\<sigma>\<^bsub>route\<^esub>(rt, dip))"
+
+lemma snd_addpre [simp]:
+  "\<And>dsn dsn' v pre. (dsn, snd(addpre (dsn', v) pre)) = addpre (dsn, v) pre"
+  unfolding addpre_def by clarsimp
+
+lemma proj2_addpreRT [simp]:
+    fixes ip rt ip' npre
+  assumes "ip\<in>kD rt"
+      and "ip'\<in>kD rt"
+    shows "\<pi>\<^sub>2(the (the (addpreRT rt ip' npre) ip)) = \<pi>\<^sub>2(the (rt ip))"
+  using assms [THEN kD_Some] unfolding addpreRT_def by clarsimp
+
+lemma proj3_addpreRT [simp]:
+    fixes ip rt ip' npre
+  assumes "ip\<in>kD rt"
+      and "ip'\<in>kD rt"
+    shows "\<pi>\<^sub>3(the (the (addpreRT rt ip' npre) ip)) = \<pi>\<^sub>3(the (rt ip))"
+  using assms [THEN kD_Some] unfolding addpreRT_def by clarsimp
+
+lemma proj5_addpreRT [simp]:
+  "\<And>rt dip ip npre. dip\<in>kD(rt) \<Longrightarrow> \<pi>\<^sub>5(the (the (addpreRT rt dip npre) ip)) = \<pi>\<^sub>5(the (rt ip))"
+  unfolding addpreRT_def by auto
+
+lemma flag_addpreRT [simp]:
+    fixes rt pre ip dip
+  assumes "dip \<in> kD rt"
+    shows "flag (the (addpreRT rt dip pre)) ip = flag rt ip"
+  unfolding addpreRT_def
+  using assms [THEN kD_Some] by (clarsimp)
+
+  lemma kD_addpreRT [simp]:
+  fixes rt dip npre
+  assumes "dip \<in> kD rt"
+  shows "kD (the (addpreRT rt dip npre)) = kD rt"
+  unfolding kD_def addpreRT_def
+  using assms [THEN kD_Some]
+  by clarsimp blast
+
+lemma vD_addpreRT [simp]:
+  fixes rt dip npre
+  assumes "dip \<in> kD rt"
+  shows "vD (the (addpreRT rt dip npre)) = vD rt"
+  unfolding vD_def addpreRT_def
+  using assms [THEN kD_Some] by clarsimp auto
+
+lemma iD_addpreRT [simp]:
+  fixes rt dip npre
+  assumes "dip \<in> kD rt"
+  shows "iD (the (addpreRT rt dip npre)) = iD rt"
+  unfolding iD_def addpreRT_def
+  using assms [THEN kD_Some] by clarsimp auto
+
+lemma nhop_addpreRT [simp]:
+    fixes rt pre ip dip
+  assumes "dip \<in> kD rt"
+    shows "nhop (the (addpreRT rt dip pre)) ip = nhop rt ip"
+  unfolding sqn_def addpreRT_def
+  using assms [THEN kD_Some] by (clarsimp)
+
+lemma sqn_addpreRT [simp]:
+    fixes rt pre ip dip
+  assumes "dip \<in> kD rt"
+    shows "sqn (the (addpreRT rt dip pre)) ip = sqn rt ip"
+  unfolding sqn_def addpreRT_def
+  using assms [THEN kD_Some] by (clarsimp)
+
+lemma dhops_addpreRT [simp]:
+    fixes rt pre ip dip
+  assumes "dip \<in> kD rt"
+    shows "dhops (the (addpreRT rt dip pre)) ip = dhops rt ip"
+  unfolding addpreRT_def
+  using assms [THEN kD_Some] by (clarsimp)
+
+lemma sqnf_addpreRT [simp]:
+  "\<And>ip dip. ip\<in>kD(rt \<xi>) \<Longrightarrow> sqnf (the (addpreRT (rt \<xi>) ip npre)) dip = sqnf (rt \<xi>) dip"
+  unfolding sqnf_def addpreRT_def by auto
+
+subsubsection "Updating route entries"
+
+lemma in_kD_case [simp]:
+    fixes dip rt
+  assumes "dip \<in> kD(rt)"
+    shows "(case rt dip of None \<Rightarrow> en | Some r \<Rightarrow> es r) = es (the (rt dip))"
+  using assms [THEN kD_Some] by auto
+
+lemma not_in_kD_case [simp]:
+    fixes dip rt
+  assumes "dip \<notin> kD(rt)"
+    shows "(case rt dip of None \<Rightarrow> en | Some r \<Rightarrow> es r) = en"
+  using assms [THEN kD_None] by auto
+
+lemma rt_Some_sqn [dest]:
+    fixes rt and ip dsn dsk flag hops nhip pre
+  assumes "rt ip = Some (dsn, dsk, flag, hops, nhip, pre)"
+    shows "sqn rt ip = dsn"
+  unfolding sqn_def using assms by simp
+
+lemma not_kD_sqn [simp]:
+    fixes dip rt
+  assumes "dip \<notin> kD(rt)"
+    shows "sqn rt dip = 0"
+  using assms unfolding sqn_def
+  by simp
+
+definition update_arg_wf :: "r \<Rightarrow> bool"
+where "update_arg_wf r \<equiv> \<pi>\<^sub>4(r) = val \<and>
+                         (\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk) \<and>
+                         (\<pi>\<^sub>3(r) = unk \<longrightarrow> \<pi>\<^sub>5(r) = 1)"
+
+lemma update_arg_wf_gives_cases:
+  "\<And>r. update_arg_wf r \<Longrightarrow> (\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)"
+  unfolding update_arg_wf_def by simp
+
+lemma update_arg_wf_tuples [simp]:
+  "\<And>nhip pre. update_arg_wf (0, unk, val, Suc 0, nhip, pre)"
+  "\<And>n hops nhip pre. update_arg_wf (Suc n, kno, val, hops,  nhip, pre)"
+  unfolding update_arg_wf_def by auto
+
+lemma update_arg_wf_tuples' [elim]:
+  "\<And>n hops nhip pre. Suc 0 \<le> n \<Longrightarrow> update_arg_wf (n, kno, val, hops,  nhip, pre)"
+  unfolding update_arg_wf_def by auto
+
+lemma wf_r_cases [intro]:
+    fixes P r
+  assumes "update_arg_wf r"
+      and c1: "\<And>nhip pre. P (0, unk, val, Suc 0, nhip, pre)"
+      and c2: "\<And>dsn hops nhip pre. dsn > 0 \<Longrightarrow> P (dsn, kno, val, hops, nhip, pre)"
+    shows "P r"
+  proof -
+    obtain dsn dsk flag hops nhip pre
+    where *: "r = (dsn, dsk, flag, hops, nhip, pre)" by (cases r)
+    with \<open>update_arg_wf r\<close> have wf1: "flag = val"
+                            and wf2: "(dsn = 0) = (dsk = unk)"
+                            and wf3: "dsk = unk \<longrightarrow> (hops = 1)"
+      unfolding update_arg_wf_def by auto
+    have "P (dsn, dsk, flag, hops, nhip, pre)"
+    proof (cases dsk)
+      assume "dsk = unk"
+      moreover with wf2 wf3 have "dsn = 0" and "hops = Suc 0" by auto
+      ultimately show ?thesis using \<open>flag = val\<close> by simp (rule c1)
+    next
+      assume "dsk = kno"
+      moreover with wf2 have "dsn > 0" by simp
+      ultimately show ?thesis using \<open>flag = val\<close> by simp (rule c2)
+    qed
+    with * show "P r" by simp
+  qed
+
+definition update :: "rt \<Rightarrow> ip \<Rightarrow> r \<Rightarrow> rt"
+  where
+  "update rt ip r \<equiv>
+     case \<sigma>\<^bsub>route\<^esub>(rt, ip) of
+       None \<Rightarrow> rt (ip \<mapsto> r)
+     | Some s \<Rightarrow>
+          if \<pi>\<^sub>2(s) < \<pi>\<^sub>2(r) then rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(s)))
+          else if \<pi>\<^sub>2(s) = \<pi>\<^sub>2(r) \<and> (\<pi>\<^sub>5(s) > \<pi>\<^sub>5(r) \<or> \<pi>\<^sub>4(s) = inv)
+               then rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(s)))
+               else if \<pi>\<^sub>3(r) = unk
+                    then rt (ip \<mapsto> (\<pi>\<^sub>2(s), snd (addpre r (\<pi>\<^sub>7(s)))))
+                    else rt (ip \<mapsto> addpre s (\<pi>\<^sub>7(r)))"
+
+lemma update_simps [simp]:
+  fixes r s nrt nr nr' ns rt ip
+  defines "s \<equiv> the \<sigma>\<^bsub>route\<^esub>(rt, ip)"
+      and "nr \<equiv> addpre r (\<pi>\<^sub>7(s))"
+      and "nr' \<equiv> (\<pi>\<^sub>2(s), \<pi>\<^sub>3(nr), \<pi>\<^sub>4(nr), \<pi>\<^sub>5(nr), \<pi>\<^sub>6(nr), \<pi>\<^sub>7(nr))"
+      and "ns \<equiv> addpre s (\<pi>\<^sub>7(r))"
+  shows
+  "\<lbrakk>ip \<notin> kD(rt)\<rbrakk>                            \<Longrightarrow> update rt ip r = rt (ip \<mapsto> r)"
+  "\<lbrakk>ip \<in> kD(rt); sqn rt ip < \<pi>\<^sub>2(r)\<rbrakk>         \<Longrightarrow> update rt ip r = rt (ip \<mapsto> nr)"
+  "\<lbrakk>ip \<in> kD(rt); sqn rt ip = \<pi>\<^sub>2(r);
+                 the (dhops rt ip) > \<pi>\<^sub>5(r)\<rbrakk> \<Longrightarrow> update rt ip r = rt (ip \<mapsto> nr)"
+  "\<lbrakk>ip \<in> kD(rt); sqn rt ip = \<pi>\<^sub>2(r);
+                 flag rt ip = Some inv\<rbrakk>     \<Longrightarrow> update rt ip r = rt (ip \<mapsto> nr)"
+  "\<lbrakk>ip \<in> kD(rt); \<pi>\<^sub>3(r) = unk; (\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)\<rbrakk>  \<Longrightarrow> update rt ip r = rt (ip \<mapsto> nr')"
+  "\<lbrakk>ip \<in> kD(rt); sqn rt ip \<ge> \<pi>\<^sub>2(r); \<pi>\<^sub>3(r) = kno;
+    sqn rt ip = \<pi>\<^sub>2(r) \<Longrightarrow> the (dhops rt ip) \<le> \<pi>\<^sub>5(r) \<and> the (flag rt ip) = val \<rbrakk>
+                                            \<Longrightarrow> update rt ip r = rt (ip \<mapsto> ns)"
+  proof -
+    assume "ip\<notin>kD(rt)"
+    hence "\<sigma>\<^bsub>route\<^esub>(rt, ip) = None" ..
+    thus "update rt ip r = rt (ip \<mapsto> r)"
+      unfolding update_def by simp
+  next
+    assume "ip \<in> kD(rt)"
+       and "sqn rt ip < \<pi>\<^sub>2(r)"
+    from this(1) obtain dsn dsk fl hops nhip pre
+      where "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)
+    with \<open>sqn rt ip < \<pi>\<^sub>2(r)\<close> show "update rt ip r = rt (ip \<mapsto> nr)"
+      unfolding update_def nr_def s_def by auto
+  next
+    assume "ip \<in> kD(rt)"
+       and "sqn rt ip = \<pi>\<^sub>2(r)"
+       and "the (dhops rt ip) > \<pi>\<^sub>5(r)"
+    from this(1) obtain dsn dsk fl hops nhip pre
+      where "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)
+    with \<open>sqn rt ip = \<pi>\<^sub>2(r)\<close> and \<open>the (dhops rt ip) > \<pi>\<^sub>5(r)\<close>
+      show "update rt ip r = rt (ip \<mapsto> nr)"
+        unfolding update_def nr_def s_def by auto
+   next
+     assume "ip \<in> kD(rt)"
+        and "sqn rt ip = \<pi>\<^sub>2(r)"
+        and "flag rt ip = Some inv"
+    from this(1) obtain dsn dsk fl hops nhip pre
+      where "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)        
+     with \<open>sqn rt ip = \<pi>\<^sub>2(r)\<close> and \<open>flag rt ip = Some inv\<close>
+      show "update rt ip r = rt (ip \<mapsto> nr)"
+        unfolding update_def nr_def s_def by auto
+   next
+    assume "ip \<in> kD(rt)"
+       and "\<pi>\<^sub>3(r) = unk"
+       and "(\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)"
+    from this(1) obtain dsn dsk fl hops nhip pre
+      where "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)
+    with \<open>(\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)\<close> and \<open>\<pi>\<^sub>3(r) = unk\<close>
+      show "update rt ip r = rt (ip \<mapsto> nr')"
+        unfolding update_def nr'_def nr_def s_def
+      by (cases r) simp
+   next
+    assume "ip \<in> kD(rt)"
+       and otherassms: "sqn rt ip \<ge> \<pi>\<^sub>2(r)"
+           "\<pi>\<^sub>3(r) = kno"
+           "sqn rt ip = \<pi>\<^sub>2(r) \<Longrightarrow> the (dhops rt ip) \<le> \<pi>\<^sub>5(r) \<and> the (flag rt ip) = val"
+    from this(1) obtain dsn dsk fl hops nhip pre
+      where "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)
+     with otherassms show "update rt ip r = rt (ip \<mapsto> ns)"
+      unfolding update_def ns_def s_def by auto
+  qed
+
+lemma update_cases [elim]:
+  assumes "(\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)"
+      and c1: "\<lbrakk>ip \<notin> kD(rt)\<rbrakk> \<Longrightarrow> P (rt (ip \<mapsto> r))"
+
+      and c2: "\<lbrakk>ip \<in> kD(rt); sqn rt ip < \<pi>\<^sub>2(r)\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c3: "\<lbrakk>ip \<in> kD(rt); sqn rt ip = \<pi>\<^sub>2(r); the (dhops rt ip) > \<pi>\<^sub>5(r)\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c4: "\<lbrakk>ip \<in> kD(rt); sqn rt ip = \<pi>\<^sub>2(r); the (flag rt ip) = inv\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c5: "\<lbrakk>ip \<in> kD(rt); \<pi>\<^sub>3(r) = unk\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> (\<pi>\<^sub>2(the \<sigma>\<^bsub>route\<^esub>(rt, ip)), \<pi>\<^sub>3(r),
+                                  \<pi>\<^sub>4(r), \<pi>\<^sub>5(r), \<pi>\<^sub>6(r), \<pi>\<^sub>7(addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))))"
+      and c6: "\<lbrakk>ip \<in> kD(rt); sqn rt ip \<ge> \<pi>\<^sub>2(r); \<pi>\<^sub>3(r) = kno;
+                sqn rt ip = \<pi>\<^sub>2(r) \<Longrightarrow> the (dhops rt ip) \<le> \<pi>\<^sub>5(r) \<and> the (flag rt ip) = val\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre (the \<sigma>\<^bsub>route\<^esub>(rt, ip)) (\<pi>\<^sub>7(r))))"
+  shows "(P (update rt ip r))"
+  proof (cases "ip \<in> kD(rt)")
+    assume "ip \<notin> kD(rt)"
+    with c1 show ?thesis
+      by simp
+  next
+    assume "ip \<in> kD(rt)"
+    moreover then obtain dsn dsk fl hops nhip pre
+      where rteq: "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)
+    moreover obtain dsn' dsk' fl' hops' nhip' pre'
+      where req: "r = (dsn', dsk', fl', hops', nhip', pre')"
+        by (cases r) metis
+    ultimately show ?thesis
+      using \<open>(\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)\<close>
+            c2 [OF \<open>ip\<in>kD(rt)\<close>]
+            c3 [OF \<open>ip\<in>kD(rt)\<close>]
+            c4 [OF \<open>ip\<in>kD(rt)\<close>]
+            c5 [OF \<open>ip\<in>kD(rt)\<close>]
+            c6 [OF \<open>ip\<in>kD(rt)\<close>]
+    unfolding update_def sqn_def by auto
+  qed
+
+lemma update_cases_kD:
+  assumes "(\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)"
+      and "ip \<in> kD(rt)"
+      and c2: "sqn rt ip < \<pi>\<^sub>2(r) \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c3: "\<lbrakk>sqn rt ip = \<pi>\<^sub>2(r); the (dhops rt ip) > \<pi>\<^sub>5(r)\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c4: "\<lbrakk>sqn rt ip = \<pi>\<^sub>2(r); the (flag rt ip) = inv\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c5: "\<pi>\<^sub>3(r) = unk \<Longrightarrow> P (rt (ip \<mapsto> (\<pi>\<^sub>2(the \<sigma>\<^bsub>route\<^esub>(rt, ip)), \<pi>\<^sub>3(r),
+                                            \<pi>\<^sub>4(r), \<pi>\<^sub>5(r), \<pi>\<^sub>6(r),
+                                            \<pi>\<^sub>7(addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))))"
+      and c6: "\<lbrakk>sqn rt ip \<ge> \<pi>\<^sub>2(r); \<pi>\<^sub>3(r) = kno;
+                sqn rt ip = \<pi>\<^sub>2(r) \<Longrightarrow> the (dhops rt ip) \<le> \<pi>\<^sub>5(r) \<and> the (flag rt ip) = val\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre (the \<sigma>\<^bsub>route\<^esub>(rt, ip)) (\<pi>\<^sub>7(r))))"
+  shows "(P (update rt ip r))"
+  using assms(1) proof (rule update_cases)
+    assume "sqn rt ip < \<pi>\<^sub>2(r)"
+    thus "P (rt(ip \<mapsto> addpre r (\<pi>\<^sub>7(the (rt ip)))))" by (rule c2)
+  next
+    assume "sqn rt ip = \<pi>\<^sub>2(r)"
+       and "the (dhops rt ip) > \<pi>\<^sub>5(r)"
+    thus "P (rt(ip \<mapsto> addpre r (\<pi>\<^sub>7 (the (rt ip)))))"
+      by (rule c3)
+  next
+    assume "sqn rt ip = \<pi>\<^sub>2(r)"
+       and "the (flag rt ip) = inv"
+    thus "P (rt(ip \<mapsto> addpre r (\<pi>\<^sub>7 (the (rt ip)))))"
+      by (rule c4)
+  next
+    assume "\<pi>\<^sub>3(r) = unk"
+    thus "P (rt (ip \<mapsto> (\<pi>\<^sub>2(the \<sigma>\<^bsub>route\<^esub>(rt, ip)), \<pi>\<^sub>3(r), \<pi>\<^sub>4(r), \<pi>\<^sub>5(r), \<pi>\<^sub>6(r),
+                        \<pi>\<^sub>7(addpre r (\<pi>\<^sub>7(the (rt ip)))))))"
+      by (rule c5)
+  next
+    assume "sqn rt ip \<ge> \<pi>\<^sub>2(r)"
+       and "\<pi>\<^sub>3(r) = kno"
+       and "sqn rt ip = \<pi>\<^sub>2(r) \<Longrightarrow> the (dhops rt ip) \<le> \<pi>\<^sub>5(r) \<and> the (flag rt ip) = val"
+    thus "P (rt (ip \<mapsto> addpre (the (rt ip)) (\<pi>\<^sub>7(r))))"
+      by (rule c6)
+  qed (simp add: \<open>ip \<in> kD(rt)\<close>)
+
+lemma in_kD_after_update [simp]:
+  fixes rt nip dsn dsk flag hops nhip pre
+  shows "kD (update rt nip (dsn, dsk, flag, hops, nhip, pre)) = insert nip (kD rt)"
+  unfolding update_def
+  by (cases "rt nip") auto
+
+lemma nhop_of_update [simp]:
+  fixes rt dip dsn dsk flag hops nhip
+  assumes "rt \<noteq> update rt dip (dsn, dsk, flag, hops, nhip, {})"
+  shows "the (nhop (update rt dip (dsn, dsk, flag, hops, nhip, {})) dip) = nhip"
+  proof -
+  from assms
+  have update_neq: "\<And>v. rt dip = Some v \<Longrightarrow>
+          update rt dip (dsn, dsk, flag, hops, nhip, {})
+             \<noteq> rt(dip \<mapsto> addpre (the (rt dip)) (\<pi>\<^sub>7 (dsn, dsk, flag, hops, nhip, {})))"
+    by auto
+  show ?thesis
+    proof (cases "rt dip = None")
+      assume "rt dip = None"
+      thus "?thesis" unfolding update_def by clarsimp
+    next
+      assume "rt dip \<noteq> None"
+      then obtain v where "rt dip = Some v" by (metis not_None_eq)
+      with update_neq [OF this] show ?thesis
+        unfolding update_def by auto
+    qed
+  qed
+
+lemma sqn_if_updated:
+  fixes rip v rt ip
+  shows "sqn (\<lambda>x. if x = rip then Some v else rt x) ip
+         = (if ip = rip then \<pi>\<^sub>2(v) else sqn rt ip)"
+  unfolding sqn_def by simp
+
+lemma update_sqn [simp]:
+  fixes rt dip rip dsn dsk hops nhip pre
+  assumes "(dsn = 0) = (dsk = unk)"
+  shows "sqn rt dip \<le> sqn (update rt rip (dsn, dsk, val, hops, nhip, pre)) dip"
+  proof (rule update_cases)
+    show "(\<pi>\<^sub>2 (dsn, dsk, val, hops, nhip, pre) = 0) = (\<pi>\<^sub>3 (dsn, dsk, val, hops, nhip, pre) = unk)"
+      by simp (rule assms)
+  qed (clarsimp simp: sqn_if_updated sqn_def)+
+
+lemma sqn_update_bigger [simp]:
+    fixes rt ip ip' dsn dsk flag hops nhip pre
+  assumes "1 \<le> hops"
+    shows "sqn rt ip \<le> sqn (update rt ip' (dsn, dsk, flag, hops, nhip, pre)) ip"
+  using assms unfolding update_def sqn_def
+  by (clarsimp split: option.split) auto
+
+lemma dhops_update [intro]:
+    fixes rt dsn dsk flag hops ip rip nhip pre
+  assumes ex: "\<forall>ip\<in>kD rt. the (dhops rt ip) \<ge> 1"
+      and ip: "(ip = rip \<and> Suc 0 \<le> hops) \<or> (ip \<noteq> rip \<and> ip\<in>kD rt)"
+    shows "Suc 0 \<le> the (dhops (update rt rip (dsn, dsk, flag, hops, nhip, pre)) ip)"
+  using ip proof
+    assume "ip = rip \<and> Suc 0 \<le> hops" thus ?thesis
+      unfolding update_def using ex
+      by (cases "rip \<in> kD rt") (drule(1) bspec, auto)
+  next
+    assume "ip \<noteq> rip \<and> ip\<in>kD rt" thus ?thesis
+      using ex unfolding update_def
+      by (cases "rip\<in>kD rt") auto
+  qed
+
+lemma update_another [simp]:
+    fixes dip ip rt dsn dsk flag hops nhip pre
+  assumes "ip \<noteq> dip"
+    shows "(update rt dip (dsn, dsk, flag, hops, nhip, pre)) ip = rt ip"
+  using assms unfolding update_def
+  by (clarsimp split: option.split)
+
+lemma nhop_update_another [simp]:
+    fixes dip ip rt dsn dsk flag hops nhip pre
+  assumes "ip \<noteq> dip"
+    shows "nhop (update rt dip (dsn, dsk, flag, hops, nhip, pre)) ip = nhop rt ip"
+  using assms unfolding update_def
+  by (clarsimp split: option.split)
+
+lemma dhops_update_another [simp]:
+    fixes dip ip rt dsn dsk flag hops nhip pre
+  assumes "ip \<noteq> dip"
+    shows "dhops (update rt dip (dsn, dsk, flag, hops, nhip, pre)) ip = dhops rt ip"
+  using assms unfolding update_def
+  by (clarsimp split: option.split)
+
+lemma sqn_update_same [simp]:
+  "\<And>rt ip dsn dsk flag hops nhip pre. sqn (rt(ip \<mapsto> v)) ip = \<pi>\<^sub>2(v)"
+  unfolding sqn_def by simp
+
+lemma dhops_update_changed [simp]:
+    fixes rt dip osn hops nhip
+  assumes "rt \<noteq> update rt dip (osn, kno, val, hops, nhip, {})"
+    shows "the (dhops (update rt dip (osn, kno, val, hops, nhip, {})) dip) = hops"
+  using assms unfolding update_def                                                      
+  by (clarsimp split: option.split_asm option.split if_split_asm) auto
+
+lemma nhop_update_unk_val [simp]:
+  "\<And>rt dip ip dsn hops npre.
+   the (nhop (update rt dip (dsn, unk, val, hops, ip, npre)) dip) = ip"
+   unfolding update_def by (clarsimp split: option.split)
+
+lemma nhop_update_changed [simp]:
+    fixes rt dip dsn dsk flg hops sip
+  assumes "update rt dip (dsn, dsk, flg, hops, sip, {}) \<noteq> rt"
+    shows "the (nhop (update rt dip (dsn, dsk, flg, hops, sip, {})) dip) = sip"
+  using assms unfolding update_def
+  by (clarsimp split: option.splits if_split_asm) auto
+
+lemma update_rt_split_asm:
+  "\<And>rt ip dsn dsk flag hops sip.
+   P (update rt ip (dsn, dsk, flag, hops, sip, {}))
+   =
+   (\<not>(rt = update rt ip (dsn, dsk, flag, hops, sip, {}) \<and> \<not>P rt
+      \<or> rt \<noteq> update rt ip (dsn, dsk, flag, hops, sip, {})
+         \<and> \<not>P (update rt ip (dsn, dsk, flag, hops, sip, {}))))"
+  by auto
+
+lemma sqn_update [simp]: "\<And>rt dip dsn flg hops sip.
+  rt \<noteq> update rt dip (dsn, kno, flg, hops, sip, {})
+  \<Longrightarrow> sqn (update rt dip (dsn, kno, flg, hops, sip, {})) dip = dsn"
+  unfolding update_def by (clarsimp split: option.split if_split_asm) auto
+
+lemma sqnf_update [simp]: "\<And>rt dip dsn dsk flg hops sip.
+  rt \<noteq> update rt dip (dsn, dsk, flg, hops, sip, {})
+  \<Longrightarrow> sqnf (update rt dip (dsn, dsk, flg, hops, sip, {})) dip = dsk"
+  unfolding update_def sqnf_def
+  by (clarsimp split: option.splits if_split_asm) auto
+
+lemma update_kno_dsn_greater_zero:
+  "\<And>rt dip ip dsn hops npre. 1 \<le> dsn \<Longrightarrow> 1 \<le> (sqn (update rt dip (dsn, kno, val, hops, ip, npre)) dip)"
+   unfolding update_def 
+  by (clarsimp split: option.splits)
+
+lemma proj3_update [simp]: "\<And>rt dip dsn dsk flg hops sip.
+  rt \<noteq> update rt dip (dsn, dsk, flg, hops, sip, {})
+  \<Longrightarrow> \<pi>\<^sub>3(the (update rt dip (dsn, dsk, flg, hops, sip, {}) dip)) = dsk"
+  unfolding update_def sqnf_def
+  by (clarsimp split: option.splits if_split_asm) auto
+
+lemma nhop_update_changed_kno_val [simp]: "\<And>rt ip dsn dsk hops nhip.
+  rt \<noteq> update rt ip (dsn, kno, val, hops, nhip, {})
+   \<Longrightarrow> the (nhop (update rt ip (dsn, kno, val, hops, nhip, {})) ip) = nhip"
+  unfolding update_def
+  by (clarsimp split: option.split_asm option.split if_split_asm) auto
+
+lemma flag_update [simp]: "\<And>rt dip dsn flg hops sip.
+  rt \<noteq> update rt dip (dsn, kno, flg, hops, sip, {})
+  \<Longrightarrow> the (flag (update rt dip (dsn, kno, flg, hops, sip, {})) dip) = flg"
+  unfolding update_def
+  by (clarsimp split: option.split if_split_asm) auto
+
+lemma the_flag_Some [dest!]:
+    fixes ip rt
+  assumes "the (flag rt ip) = x"
+      and "ip \<in> kD rt"
+    shows "flag rt ip = Some x"
+  using assms by auto
+
+lemma kD_update_unchanged [dest]:
+    fixes rt dip dsn dsk flag hops nhip pre
+  assumes "rt = update rt dip (dsn, dsk, flag, hops, nhip, pre)"
+    shows "dip\<in>kD(rt)"
+  proof -
+    have "dip\<in>kD(update rt dip (dsn, dsk, flag, hops, nhip, pre))" by simp
+    with assms show ?thesis by simp
+  qed
+
+lemma nhop_update [simp]: "\<And>rt dip dsn dsk flg hops sip.
+  rt \<noteq> update rt dip (dsn, dsk, flg, hops, sip, {})
+  \<Longrightarrow> the (nhop (update rt dip (dsn, dsk, flg, hops, sip, {})) dip) = sip"
+  unfolding update_def sqnf_def
+  by (clarsimp split: option.splits if_split_asm) auto
+
+lemma sqn_update_another [simp]:
+    fixes dip ip rt dsn dsk flag hops nhip pre
+  assumes "ip \<noteq> dip"
+    shows "sqn (update rt dip (dsn, dsk, flag, hops, nhip, pre)) ip = sqn rt ip"
+  using assms unfolding update_def sqn_def
+  by (clarsimp split: option.splits) auto
+
+lemma sqnf_update_another [simp]:
+    fixes dip ip rt dsn dsk flag hops nhip pre
+  assumes "ip \<noteq> dip"
+    shows "sqnf (update rt dip (dsn, dsk, flag, hops, nhip, pre)) ip = sqnf rt ip"
+  using assms unfolding update_def sqnf_def
+  by (clarsimp split: option.splits) auto
+
+lemma vD_update_val [dest]:
+  "\<And>dip rt dip' dsn dsk hops nhip pre.
+   dip \<in> vD(update rt dip' (dsn, dsk, val, hops, nhip, pre)) \<Longrightarrow> (dip\<in>vD(rt) \<or> dip=dip')"
+   unfolding update_def vD_def by (clarsimp split: option.split_asm if_split_asm)
+
+subsubsection "Invalidating route entries"
+
+definition invalidate :: "rt \<Rightarrow> (ip \<rightharpoonup> sqn) \<Rightarrow> rt"
+where "invalidate rt dests \<equiv>
+  \<lambda>ip. case (rt ip, dests ip) of
+    (None, _) \<Rightarrow> None
+  | (Some s, None) \<Rightarrow> Some s
+  | (Some (_, dsk, _, hops, nhip, pre), Some rsn) \<Rightarrow>
+                      Some (rsn, dsk, inv, hops, nhip, pre)"
+
+lemma proj3_invalidate [simp]:
+  "\<And>dip. \<pi>\<^sub>3(the ((invalidate rt dests) dip)) = \<pi>\<^sub>3(the (rt dip))"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+lemma proj5_invalidate [simp]:
+  "\<And>dip. \<pi>\<^sub>5(the ((invalidate rt dests) dip)) = \<pi>\<^sub>5(the (rt dip))"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+lemma proj6_invalidate [simp]:
+  "\<And>dip. \<pi>\<^sub>6(the ((invalidate rt dests) dip)) = \<pi>\<^sub>6(the (rt dip))"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+lemma proj7_invalidate [simp]:
+  "\<And>dip. \<pi>\<^sub>7(the ((invalidate rt dests) dip)) = \<pi>\<^sub>7(the (rt dip))"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+lemma invalidate_kD_inv [simp]:
+  "\<And>rt dests. kD (invalidate rt dests) = kD rt"
+  unfolding invalidate_def kD_def
+  by (simp split: option.split)
+
+lemma invalidate_sqn:
+  fixes rt dip dests
+  assumes "\<forall>rsn. dests dip = Some rsn \<longrightarrow> sqn rt dip \<le> rsn"
+  shows "sqn rt dip \<le> sqn (invalidate rt dests) dip"
+  proof (cases "dip \<notin> kD(rt)")
+    assume "\<not> dip \<notin> kD(rt)"
+    hence "dip\<in>kD(rt)" by simp
+    then obtain dsn dsk flag hops nhip pre where "rt dip = Some (dsn, dsk, flag, hops, nhip, pre)"
+      by (metis kD_Some)
+    with assms show "sqn rt dip \<le> sqn (invalidate rt dests) dip"
+      by (cases "dests dip") (auto simp add: invalidate_def sqn_def)
+  qed simp
+
+lemma sqn_invalidate_in_dests [simp]:
+    fixes dests ipa rsn rt
+  assumes "dests ipa = Some rsn"
+      and "ipa\<in>kD(rt)"
+    shows "sqn (invalidate rt dests) ipa = rsn"
+  unfolding invalidate_def sqn_def
+  using assms(1) assms(2) [THEN kD_Some]
+  by clarsimp
+
+lemma dhops_invalidate [simp]:
+  "\<And>dip. the (dhops (invalidate rt dests) dip) = the (dhops rt dip)"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+lemma sqnf_invalidate [simp]:
+  "\<And>dip. sqnf (invalidate (rt \<xi>) (dests \<xi>)) dip = sqnf (rt \<xi>) dip"
+  unfolding sqnf_def invalidate_def by (clarsimp split: option.split)
+
+lemma nhop_invalidate [simp]:
+  "\<And>dip. the (nhop (invalidate (rt \<xi>) (dests \<xi>)) dip) = the (nhop (rt \<xi>) dip)"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+lemma invalidate_other [simp]:
+    fixes rt dests dip
+  assumes "dip\<notin>dom(dests)"
+    shows "invalidate rt dests dip = rt dip"
+  using assms unfolding invalidate_def
+  by (clarsimp split: option.split_asm)
+
+lemma invalidate_none [simp]:
+    fixes rt dests dip
+  assumes "dip\<notin>kD(rt)"
+    shows "invalidate rt dests dip = None"
+  using assms unfolding invalidate_def by clarsimp
+
+lemma vD_invalidate_vD_not_dests:
+  "\<And>dip rt dests. dip\<in>vD(invalidate rt dests) \<Longrightarrow> dip\<in>vD(rt) \<and> dests dip = None"
+  unfolding invalidate_def vD_def 
+  by (clarsimp split: option.split_asm)
+
+lemma sqn_invalidate_not_in_dests [simp]:
+  fixes dests dip rt
+  assumes "dip\<notin>dom(dests)"
+  shows "sqn (invalidate rt dests) dip = sqn rt dip"
+  using assms unfolding sqn_def by simp
+
+lemma invalidate_changes:
+    fixes rt dests dip dsn dsk flag hops nhip pre
+  assumes "invalidate rt dests dip = Some (dsn, dsk, flag, hops, nhip, pre)"
+    shows "  dsn = (case dests dip of None \<Rightarrow> \<pi>\<^sub>2(the (rt dip)) | Some rsn \<Rightarrow> rsn)
+           \<and> dsk = \<pi>\<^sub>3(the (rt dip))
+           \<and> flag = (if dests dip = None then \<pi>\<^sub>4(the (rt dip)) else inv)
+           \<and> hops = \<pi>\<^sub>5(the (rt dip))
+           \<and> nhip = \<pi>\<^sub>6(the (rt dip))
+           \<and> pre = \<pi>\<^sub>7(the (rt dip))"
+  using assms unfolding invalidate_def
+  by (cases "rt dip", clarsimp, cases "dests dip") auto
+
+
+lemma proj3_inv: "\<And>dip rt dests. dip\<in>kD (rt)
+                      \<Longrightarrow> \<pi>\<^sub>3(the (invalidate rt dests dip)) = \<pi>\<^sub>3(the (rt dip))"
+  by (clarsimp simp: invalidate_def kD_def split: option.split)
+
+lemma dests_iD_invalidate [simp]:
+  assumes "dests ip = Some rsn"
+      and "ip\<in>kD(rt)"
+    shows "ip\<in>iD(invalidate rt dests)"
+  using assms(1) assms(2) [THEN kD_Some] unfolding invalidate_def iD_def
+  by (clarsimp split: option.split)
+
+subsection "Route Requests"
+
+text \<open>Generate a fresh route request identifier.\<close>
+
+definition nrreqid :: "(ip \<times> rreqid) set \<Rightarrow> ip \<Rightarrow> rreqid"
+  where "nrreqid rreqs ip \<equiv> Max ({n. (ip, n) \<in> rreqs} \<union> {0}) + 1"
+
+subsection "Queued Packets"
+
+text \<open>Functions for sending data packets.\<close>
+
+type_synonym store = "ip \<rightharpoonup> (p \<times> data list)"
+
+definition sigma_queue :: "store \<Rightarrow> ip \<Rightarrow> data list"    ("\<sigma>\<^bsub>queue\<^esub>'(_, _')")
+  where "\<sigma>\<^bsub>queue\<^esub>(store, dip) \<equiv> case store dip of None \<Rightarrow> [] | Some (p, q) \<Rightarrow> q"
+
+definition qD :: "store \<Rightarrow> ip set"
+  where "qD \<equiv> dom"
+
+definition add :: "data \<Rightarrow> ip \<Rightarrow> store \<Rightarrow> store"
+  where "add d dip store \<equiv> case store dip of
+                              None \<Rightarrow> store (dip \<mapsto> (req, [d]))
+                            | Some (p, q) \<Rightarrow> store (dip \<mapsto> (p, q @ [d]))"
+
+lemma qD_add [simp]:
+  fixes d dip store
+  shows "qD(add d dip store) = insert dip (qD store)"
+  unfolding add_def Let_def qD_def
+  by (clarsimp split: option.split)
+
+definition drop :: "ip \<Rightarrow> store \<rightharpoonup> store"
+  where "drop dip store \<equiv>
+    map_option (\<lambda>(p, q). if tl q = [] then store (dip := None)
+                                      else store (dip \<mapsto> (p, tl q))) (store dip)"
+
+definition sigma_p_flag :: "store \<Rightarrow> ip \<rightharpoonup> p" ("\<sigma>\<^bsub>p-flag\<^esub>'(_, _')")
+  where "\<sigma>\<^bsub>p-flag\<^esub>(store, dip) \<equiv> map_option fst (store dip)"
+
+definition unsetRRF :: "store \<Rightarrow> ip \<Rightarrow> store"
+  where "unsetRRF store dip \<equiv> case store dip of
+                                None \<Rightarrow> store
+                              | Some (p, q) \<Rightarrow> store (dip \<mapsto> (noreq, q))"
+
+definition setRRF :: "store \<Rightarrow> (ip \<rightharpoonup> sqn) \<Rightarrow> store"
+  where "setRRF store dests \<equiv> \<lambda>dip. if dests dip = None then store dip
+                                    else map_option (\<lambda>(_, q). (req, q)) (store dip)"
+
+subsection "Comparison with the original technical report"
+
+text \<open>
+  The major differences with the AODV technical report of Fehnker et al are:
+  \begin{enumerate}
+  \item @{term nhop} is partial, thus a `@{term the}' is needed, similarly for @{term dhops}
+        and @{term addpreRT}.
+  \item @{term precs} is partial.
+  \item @{term "\<sigma>\<^bsub>p-flag\<^esub>(store, dip)"} is partial.
+  \item The routing table (@{typ rt}) is modelled as a map (@{typ "ip \<Rightarrow> r option"})
+        rather than a set of 7-tuples, likewise, the @{typ r} is a 6-tuple rather than
+        a 7-tuple, i.e., the destination ip-address (@{term "dip"}) is taken from the
+        argument to the function, rather than a part of the result. Well-definedness then
+        follows from the structure of the type and more related facts are available
+        automatically, rather than having to be acquired through tedious proofs.
+  \item Similar remarks hold for the dests mapping passed to @{term "invalidate"},
+        and @{term "store"}.
+  \end{enumerate}
+\<close>
+
+end
+

formal/afp/AODV/Aodv_Loop_Freedom.thy

@@ -0,0 +1,369 @@
+(*  Title:       Aodv_Loop_Freedom.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "Lift and transfer invariants to show loop freedom"
+
+theory Aodv_Loop_Freedom
+imports AWN.OClosed_Transfer AWN.Qmsg_Lifting Global_Invariants Loop_Freedom
+begin
+
+subsection \<open>Lift to parallel processes with queues\<close>
+
+lemma par_step_no_change_on_send_or_receive:
+    fixes \<sigma> s a \<sigma>' s'
+  assumes "((\<sigma>, s), a, (\<sigma>', s')) \<in> oparp_sos i (oseqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i) (seqp_sos \<Gamma>\<^sub>Q\<^sub>M\<^sub>S\<^sub>G)"
+      and "a \<noteq> \<tau>"
+    shows "\<sigma>' i = \<sigma> i"
+  using assms by (rule qmsg_no_change_on_send_or_receive)
+
+lemma par_nhop_quality_increases:
+  shows "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile> (otherwith ((=)) {i} (orecvmsg (\<lambda>\<sigma> m.
+                                    msg_fresh \<sigma> m \<and> msg_zhops m)),
+                                  other quality_increases {i} \<rightarrow>)
+                        global (\<lambda>\<sigma>. \<forall>dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                          in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                                             \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+    proof (rule lift_into_qmsg [OF seq_nhop_quality_increases])
+    show "opaodv i \<Turnstile>\<^sub>A (otherwith ((=)) {i}
+                         (orecvmsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)),
+                        other quality_increases {i} \<rightarrow>)
+                       globala (\<lambda>(\<sigma>, _, \<sigma>'). quality_increases (\<sigma> i) (\<sigma>' i))"
+    proof (rule ostep_invariant_weakenE [OF oquality_increases], simp_all)
+      fix t :: "(((nat \<Rightarrow> state) \<times> (state, msg, pseqp, pseqp label) seqp), msg seq_action) transition"
+      assume "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<sigma>, _), _, (\<sigma>', _)). \<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)) t"
+      thus "quality_increases (fst (fst t) i) (fst (snd (snd t)) i)"
+        by (cases t) (clarsimp dest!: onllD, metis aodv_ex_label)
+    next
+      fix \<sigma> \<sigma>' a
+      assume "otherwith ((=)) {i}
+                (orecvmsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)) \<sigma> \<sigma>' a"
+      thus "otherwith quality_increases {i} (orecvmsg (\<lambda>_. rreq_rrep_sn)) \<sigma> \<sigma>' a"
+        by - (erule weaken_otherwith, auto)
+    qed
+  qed auto
+
+lemma par_rreq_rrep_sn_quality_increases:
+  "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                            globala (\<lambda>(\<sigma>, _, \<sigma>'). quality_increases (\<sigma> i) (\<sigma>' i))"
+  proof -
+    have "opaodv i \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                globala (\<lambda>(\<sigma>, _, \<sigma>'). quality_increases (\<sigma> i) (\<sigma>' i))"
+      by (rule ostep_invariant_weakenE [OF olocal_quality_increases])
+         (auto dest!: onllD seqllD elim!: aodv_ex_labelE)
+    hence "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                                   globala (\<lambda>(\<sigma>, _, \<sigma>'). quality_increases (\<sigma> i) (\<sigma>' i))"
+      by (rule lift_step_into_qmsg_statelessassm) simp_all
+    thus ?thesis by rule auto
+  qed
+
+lemma par_rreq_rrep_nsqn_fresh_any_step:
+  shows "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>,
+                                   other (\<lambda>_ _. True) {i} \<rightarrow>)
+                                  globala (\<lambda>(\<sigma>, a, \<sigma>'). anycast (msg_fresh \<sigma>) a)"
+  proof -
+    have "opaodv i \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. (orecvmsg (\<lambda>_. rreq_rrep_sn)) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                       globala (\<lambda>(\<sigma>, a, \<sigma>'). anycast (msg_fresh \<sigma>) a)"
+    proof (rule ostep_invariant_weakenE [OF rreq_rrep_nsqn_fresh_any_step_invariant])
+      fix t
+      assume "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<sigma>, _), a, _). anycast (msg_fresh \<sigma>) a) t"
+      thus "globala (\<lambda>(\<sigma>, a, \<sigma>'). anycast (msg_fresh \<sigma>) a) t"
+        by (cases t) (clarsimp dest!: onllD, metis aodv_ex_label)
+    qed auto
+    hence "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. (orecvmsg (\<lambda>_. rreq_rrep_sn)) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                                    globala (\<lambda>(\<sigma>, a, \<sigma>'). anycast (msg_fresh \<sigma>) a)"
+      by (rule lift_step_into_qmsg_statelessassm) simp_all
+    thus ?thesis by rule auto
+  qed
+
+lemma par_anycast_msg_zhops:
+  shows "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                                  globala (\<lambda>(_, a, _). anycast msg_zhops a)"
+  proof -
+    from anycast_msg_zhops initiali_aodv oaodv_trans aodv_trans
+      have "opaodv i \<Turnstile>\<^sub>A (act TT, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                         seqll i (onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(_, a, _). anycast msg_zhops a))"
+        by (rule open_seq_step_invariant)
+    hence "opaodv i \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                            globala (\<lambda>(_, a, _). anycast msg_zhops a)"
+    proof (rule ostep_invariant_weakenE)
+      fix t :: "(((nat \<Rightarrow> state) \<times> (state, msg, pseqp, pseqp label) seqp), msg seq_action) transition"
+      assume "seqll i (onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(_, a, _). anycast msg_zhops a)) t"
+      thus "globala (\<lambda>(_, a, _). anycast msg_zhops a) t"
+        by (cases t) (clarsimp dest!: seqllD onllD, metis aodv_ex_label)
+    qed simp_all
+    hence "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                                     globala (\<lambda>(_, a, _). anycast msg_zhops a)"
+      by (rule lift_step_into_qmsg_statelessassm) simp_all
+    thus ?thesis by rule auto
+  qed
+
+subsection \<open>Lift to nodes\<close>
+
+lemma node_step_no_change_on_send_or_receive:
+  assumes "((\<sigma>, NodeS i P R), a, (\<sigma>', NodeS i' P' R')) \<in> onode_sos
+                                      (oparp_sos i (oseqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i) (seqp_sos \<Gamma>\<^sub>Q\<^sub>M\<^sub>S\<^sub>G))"
+      and "a \<noteq> \<tau>"
+    shows "\<sigma>' i = \<sigma> i"
+  using assms
+  by (cases a) (auto elim!: par_step_no_change_on_send_or_receive)
+
+lemma node_nhop_quality_increases:
+  shows "\<langle> i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R \<rangle>\<^sub>o \<Turnstile>
+           (otherwith ((=)) {i}
+              (oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)),
+              other quality_increases {i}
+            \<rightarrow>) global (\<lambda>\<sigma>. \<forall>dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                  in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                                     \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+  by (rule node_lift [OF par_nhop_quality_increases]) auto
+
+lemma node_quality_increases:
+  "\<langle> i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R \<rangle>\<^sub>o \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma>,
+                                         other (\<lambda>_ _. True) {i} \<rightarrow>)
+                            globala (\<lambda>(\<sigma>, _, \<sigma>'). quality_increases (\<sigma> i) (\<sigma>' i))"
+  by (rule node_lift_step_statelessassm [OF par_rreq_rrep_sn_quality_increases]) simp
+
+lemma node_rreq_rrep_nsqn_fresh_any_step:
+  shows "\<langle> i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R \<rangle>\<^sub>o \<Turnstile>\<^sub>A
+          (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+          globala (\<lambda>(\<sigma>, a, \<sigma>'). castmsg (msg_fresh \<sigma>) a)"
+  by (rule node_lift_anycast_statelessassm [OF par_rreq_rrep_nsqn_fresh_any_step])
+
+lemma node_anycast_msg_zhops:
+  shows "\<langle> i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R \<rangle>\<^sub>o \<Turnstile>\<^sub>A
+          (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+          globala (\<lambda>(_, a, _). castmsg msg_zhops a)"
+  by (rule node_lift_anycast_statelessassm [OF par_anycast_msg_zhops])
+
+lemma node_silent_change_only:
+  shows "\<langle> i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<^sub>i \<rangle>\<^sub>o \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_ _. True) \<sigma>,
+                                               other (\<lambda>_ _. True) {i} \<rightarrow>)
+          globala (\<lambda>(\<sigma>, a, \<sigma>'). a \<noteq> \<tau> \<longrightarrow> \<sigma>' i = \<sigma> i)"
+  proof (rule ostep_invariantI, simp (no_asm), rule impI)
+    fix \<sigma> \<zeta> a \<sigma>' \<zeta>'
+    assume or: "(\<sigma>, \<zeta>) \<in> oreachable (\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<^sub>i\<rangle>\<^sub>o)
+                                    (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_ _. True) \<sigma>)
+                                    (other (\<lambda>_ _. True) {i})"
+      and tr: "((\<sigma>, \<zeta>), a, (\<sigma>', \<zeta>')) \<in> trans (\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<^sub>i\<rangle>\<^sub>o)"
+      and "a \<noteq> \<tau>\<^sub>n"
+    from or obtain p R where "\<zeta> = NodeS i p R"
+      by - (drule node_net_state, metis)
+    with tr have "((\<sigma>, NodeS i p R), a, (\<sigma>', \<zeta>'))
+                     \<in> onode_sos (oparp_sos i (trans (opaodv i)) (trans qmsg))"
+      by simp
+    thus "\<sigma>' i = \<sigma> i" using \<open>a \<noteq> \<tau>\<^sub>n\<close> 
+      by (cases rule: onode_sos.cases)
+         (auto elim: qmsg_no_change_on_send_or_receive)
+  qed
+
+subsection \<open>Lift to partial networks\<close>
+
+lemma arrive_rreq_rrep_nsqn_fresh_inc_sn [simp]:
+  assumes "oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> P \<sigma> m) \<sigma> m"
+    shows "oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma> m"
+  using assms by (cases m) auto
+
+lemma opnet_nhop_quality_increases:
+  shows "opnet (\<lambda>i. opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg) p \<Turnstile>
+           (otherwith ((=)) (net_tree_ips p)
+              (oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)),
+               other quality_increases (net_tree_ips p) \<rightarrow>)
+              global (\<lambda>\<sigma>. \<forall>i\<in>net_tree_ips p. \<forall>dip.
+                          let nhip = the (nhop (rt (\<sigma> i)) dip)
+                          in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                             \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+  proof (rule pnet_lift [OF node_nhop_quality_increases])
+    fix i R
+    have "\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma>,
+                                              other (\<lambda>_ _. True) {i} \<rightarrow>) globala (\<lambda>(\<sigma>, a, \<sigma>').
+            castmsg (\<lambda>m. msg_fresh \<sigma> m \<and> msg_zhops m) a)"
+    proof (rule ostep_invariantI, simp (no_asm))
+      fix \<sigma> s a \<sigma>' s'
+      assume or: "(\<sigma>, s) \<in> oreachable (\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o)
+                                      (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma>)
+                                      (other (\<lambda>_ _. True) {i})"
+         and tr: "((\<sigma>, s), a, (\<sigma>', s')) \<in> trans (\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o)"
+         and am: "oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma> a"
+      from or tr am have "castmsg (msg_fresh \<sigma>) a"
+        by (auto dest!: ostep_invariantD [OF node_rreq_rrep_nsqn_fresh_any_step])
+      moreover from or tr am have "castmsg (msg_zhops) a"
+        by (auto dest!: ostep_invariantD [OF node_anycast_msg_zhops])
+      ultimately show "castmsg (\<lambda>m. msg_fresh \<sigma> m \<and> msg_zhops m) a"
+        by (case_tac a) auto
+    qed
+    thus "\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o \<Turnstile>\<^sub>A
+            (\<lambda>\<sigma> _. oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m) \<sigma>,
+             other quality_increases {i} \<rightarrow>) globala (\<lambda>(\<sigma>, a, _).
+               castmsg (\<lambda>m. msg_fresh \<sigma> m \<and> msg_zhops m) a)"
+      by rule auto
+  next
+    fix i R
+    show "\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o \<Turnstile>\<^sub>A
+            (\<lambda>\<sigma> _. oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m) \<sigma>,
+             other quality_increases {i} \<rightarrow>) globala (\<lambda>(\<sigma>, a, \<sigma>').
+               a \<noteq> \<tau> \<and> (\<forall>d. a \<noteq> i:deliver(d)) \<longrightarrow> \<sigma> i = \<sigma>' i)"
+      by (rule ostep_invariant_weakenE [OF node_silent_change_only]) auto
+  next
+    fix i R
+    show "\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o \<Turnstile>\<^sub>A
+            (\<lambda>\<sigma> _. oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m) \<sigma>,
+             other quality_increases {i} \<rightarrow>) globala (\<lambda>(\<sigma>, a, \<sigma>').
+               a = \<tau> \<or> (\<exists>d. a = i:deliver(d)) \<longrightarrow> quality_increases (\<sigma> i) (\<sigma>' i))"
+      by (rule ostep_invariant_weakenE [OF node_quality_increases]) auto
+  qed simp_all
+
+subsection \<open>Lift to closed networks\<close>
+
+lemma onet_nhop_quality_increases:
+  shows "oclosed (opnet (\<lambda>i. opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg) p)
+           \<Turnstile> (\<lambda>_ _ _. True, other quality_increases (net_tree_ips p) \<rightarrow>)
+              global (\<lambda>\<sigma>. \<forall>i\<in>net_tree_ips p. \<forall>dip.
+                          let nhip = the (nhop (rt (\<sigma> i)) dip)
+                          in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                             \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+  (is "_ \<Turnstile> (_, ?U \<rightarrow>) ?inv")
+  proof (rule inclosed_closed)
+    from opnet_nhop_quality_increases
+      show "opnet (\<lambda>i. opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg) p
+               \<Turnstile> (otherwith ((=)) (net_tree_ips p) inoclosed, ?U \<rightarrow>) ?inv"
+    proof (rule oinvariant_weakenE)
+      fix \<sigma> \<sigma>' :: "ip \<Rightarrow> state" and a :: "msg node_action"
+      assume "otherwith ((=)) (net_tree_ips p) inoclosed \<sigma> \<sigma>' a"
+      thus "otherwith ((=)) (net_tree_ips p)
+              (oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)) \<sigma> \<sigma>' a"
+      proof (rule otherwithEI)
+        fix \<sigma> :: "ip \<Rightarrow> state" and a :: "msg node_action"
+        assume "inoclosed \<sigma> a"
+        thus "oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m) \<sigma> a"
+        proof (cases a)
+          fix ii ni ms
+          assume "a = ii\<not>ni:arrive(ms)"
+          moreover with \<open>inoclosed \<sigma> a\<close> obtain d di where "ms = newpkt(d, di)"
+            by (cases ms) auto
+          ultimately show ?thesis by simp
+        qed simp_all
+      qed
+    qed
+  qed
+
+subsection \<open>Transfer into the standard model\<close>
+
+interpretation aodv_openproc: openproc paodv opaodv id
+  rewrites "aodv_openproc.initmissing = initmissing"
+  proof -
+    show "openproc paodv opaodv id"
+    proof unfold_locales
+      fix i :: ip
+      have "{(\<sigma>, \<zeta>). (\<sigma> i, \<zeta>) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i \<and> (\<forall>j. j \<noteq> i \<longrightarrow> \<sigma> j \<in> fst ` \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V j)} \<subseteq> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'"
+        unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'_def
+        proof (rule equalityD1)
+          show "\<And>f p. {(\<sigma>, \<zeta>). (\<sigma> i, \<zeta>) \<in> {(f i, p)} \<and> (\<forall>j. j \<noteq> i
+                      \<longrightarrow> \<sigma> j \<in> fst ` {(f j, p)})} = {(f, p)}"
+            by (rule set_eqI) auto
+        qed
+      thus "{ (\<sigma>, \<zeta>) |\<sigma> \<zeta> s. s \<in> init (paodv i)
+                             \<and> (\<sigma> i, \<zeta>) = id s
+                             \<and> (\<forall>j. j\<noteq>i \<longrightarrow> \<sigma> j \<in> (fst o id) ` init (paodv j)) } \<subseteq> init (opaodv i)"
+        by simp
+    next
+      show "\<forall>j. init (paodv j) \<noteq> {}"
+        unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by simp
+    next
+      fix i s a s' \<sigma> \<sigma>'
+      assume "\<sigma> i = fst (id s)"
+         and "\<sigma>' i = fst (id s')"
+         and "(s, a, s') \<in> trans (paodv i)"
+      then obtain q q' where "s = (\<sigma> i, q)"
+                         and "s' = (\<sigma>' i, q')"
+                         and "((\<sigma> i, q), a, (\<sigma>' i, q')) \<in> trans (paodv i)" 
+         by (cases s, cases s') auto
+      from this(3) have "((\<sigma>, q), a, (\<sigma>', q')) \<in> trans (opaodv i)"
+        by simp (rule open_seqp_action [OF aodv_wf])
+
+      with \<open>s = (\<sigma> i, q)\<close> and \<open>s' = (\<sigma>' i, q')\<close>
+        show "((\<sigma>, snd (id s)), a, (\<sigma>', snd (id s'))) \<in> trans (opaodv i)"
+          by simp
+    qed
+    then interpret opn: openproc paodv opaodv id .
+    have [simp]: "\<And>i. (SOME x. x \<in> (fst o id) ` init (paodv i)) = aodv_init i"
+      unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by simp
+    hence "\<And>i. openproc.initmissing paodv id i = initmissing i"
+      unfolding opn.initmissing_def opn.someinit_def initmissing_def
+      by (auto split: option.split)
+    thus "openproc.initmissing paodv id = initmissing" ..
+  qed
+
+interpretation aodv_openproc_par_qmsg: openproc_parq paodv opaodv id qmsg
+  rewrites "aodv_openproc_par_qmsg.netglobal = netglobal"
+    and "aodv_openproc_par_qmsg.initmissing = initmissing"
+  proof -
+    show "openproc_parq paodv opaodv id qmsg"
+      by (unfold_locales) simp
+    then interpret opq: openproc_parq paodv opaodv id qmsg .
+
+    have im: "\<And>\<sigma>. openproc.initmissing (\<lambda>i. paodv i \<langle>\<langle> qmsg) (\<lambda>(p, q). (fst (id p), snd (id p), q)) \<sigma>
+                                                                                    = initmissing \<sigma>"
+      unfolding opq.initmissing_def opq.someinit_def initmissing_def
+      unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def \<sigma>\<^sub>Q\<^sub>M\<^sub>S\<^sub>G_def by (clarsimp cong: option.case_cong)
+    thus "openproc.initmissing (\<lambda>i. paodv i \<langle>\<langle> qmsg) (\<lambda>(p, q). (fst (id p), snd (id p), q)) = initmissing"
+      by (rule ext)
+    have "\<And>P \<sigma>. openproc.netglobal (\<lambda>i. paodv i \<langle>\<langle> qmsg) (\<lambda>(p, q). (fst (id p), snd (id p), q)) P \<sigma>
+                                                                                = netglobal P \<sigma>"
+      unfolding opq.netglobal_def netglobal_def opq.initmissing_def initmissing_def opq.someinit_def
+      unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def \<sigma>\<^sub>Q\<^sub>M\<^sub>S\<^sub>G_def
+      by (clarsimp cong: option.case_cong
+                   simp del: One_nat_def
+                   simp add: fst_initmissing_netgmap_default_aodv_init_netlift
+                                                  [symmetric, unfolded initmissing_def])
+    thus "openproc.netglobal (\<lambda>i. paodv i \<langle>\<langle> qmsg) (\<lambda>(p, q). (fst (id p), snd (id p), q)) = netglobal"
+      by auto
+  qed
+
+lemma net_nhop_quality_increases:
+  assumes "wf_net_tree n"
+  shows "closed (pnet (\<lambda>i. paodv i \<langle>\<langle> qmsg) n) \<TTurnstile> netglobal
+                           (\<lambda>\<sigma>. \<forall>i dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                        in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                                            \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+        (is "_ \<TTurnstile> netglobal (\<lambda>\<sigma>. \<forall>i. ?inv \<sigma> i)")
+  proof -
+    from \<open>wf_net_tree n\<close>
+      have proto: "closed (pnet (\<lambda>i. paodv i \<langle>\<langle> qmsg) n) \<TTurnstile> netglobal (\<lambda>\<sigma>. \<forall>i\<in>net_tree_ips n. \<forall>dip.
+                                            let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                            in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                                                \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+        by (rule aodv_openproc_par_qmsg.close_opnet [OF _ onet_nhop_quality_increases])
+    show ?thesis
+    unfolding invariant_def opnet_sos.opnet_tau1
+    proof (rule, simp only: aodv_openproc_par_qmsg.netglobalsimp
+                            fst_initmissing_netgmap_pair_fst, rule allI)
+      fix \<sigma> i
+      assume sr: "\<sigma> \<in> reachable (closed (pnet (\<lambda>i. paodv i \<langle>\<langle> qmsg) n)) TT"
+      hence "\<forall>i\<in>net_tree_ips n. ?inv (fst (initmissing (netgmap fst \<sigma>))) i"
+        by - (drule invariantD [OF proto],
+              simp only: aodv_openproc_par_qmsg.netglobalsimp
+                         fst_initmissing_netgmap_pair_fst)
+      thus "?inv (fst (initmissing (netgmap fst \<sigma>))) i"
+      proof (cases "i\<in>net_tree_ips n")
+        assume "i\<notin>net_tree_ips n"
+        from sr have "\<sigma> \<in> reachable (pnet (\<lambda>i. paodv i \<langle>\<langle> qmsg) n) TT" ..
+        hence "net_ips \<sigma> = net_tree_ips n" ..
+        with \<open>i\<notin>net_tree_ips n\<close> have "i\<notin>net_ips \<sigma>" by simp
+        hence "(fst (initmissing (netgmap fst \<sigma>))) i = aodv_init i"
+          by simp
+        thus ?thesis by simp
+      qed metis
+    qed
+  qed
+
+subsection \<open>Loop freedom of AODV\<close>
+
+theorem aodv_loop_freedom:
+  assumes "wf_net_tree n"
+  shows "closed (pnet (\<lambda>i. paodv i \<langle>\<langle> qmsg) n) \<TTurnstile> netglobal (\<lambda>\<sigma>. \<forall>dip. irrefl ((rt_graph \<sigma> dip)\<^sup>+))"
+  using assms by (rule aodv_openproc_par_qmsg.netglobal_weakenE
+                          [OF net_nhop_quality_increases inv_to_loop_freedom])
+
+end

formal/afp/AODV/Aodv_Message.thy

@@ -0,0 +1,74 @@
+(*  Title:       Aodv_Message.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "AODV protocol messages"
+
+theory Aodv_Message
+imports Aodv_Basic
+begin
+
+datatype msg =
+    Rreq nat rreqid ip sqn k ip sqn ip
+  | Rrep nat ip sqn ip ip
+  | Rerr "ip \<rightharpoonup> sqn" ip
+  | Newpkt data ip
+  | Pkt data ip ip
+
+instantiation msg :: msg
+begin
+  definition newpkt_def [simp]: "newpkt \<equiv> \<lambda>(d, dip). Newpkt d dip"
+  definition eq_newpkt_def: "eq_newpkt m \<equiv> case m of Newpkt d dip \<Rightarrow> True | _ \<Rightarrow> False"
+
+  instance by intro_classes (simp add: eq_newpkt_def)
+end
+
+text \<open>The @{type msg} type models the different messages used within AODV.
+      The instantiation as a @{class msg} is a technicality due to the special
+      treatment of @{term newpkt} messages in the AWN SOS rules.
+      This use of classes allows a clean separation of the AWN-specific definitions
+      and these AODV-specific definitions.\<close>
+
+definition rreq :: "nat \<times> rreqid \<times> ip \<times> sqn \<times> k \<times> ip \<times> sqn \<times> ip \<Rightarrow> msg"
+  where "rreq \<equiv> \<lambda>(hops, rreqid, dip, dsn, dsk, oip, osn, sip).
+                    Rreq hops rreqid dip dsn dsk oip osn sip"
+
+lemma rreq_simp [simp]:
+  "rreq(hops, rreqid, dip, dsn, dsk, oip, osn, sip) =  Rreq hops rreqid dip dsn dsk oip osn sip"
+  unfolding rreq_def by simp
+
+definition rrep :: "nat \<times> ip \<times> sqn \<times> ip \<times> ip \<Rightarrow> msg"
+  where "rrep \<equiv> \<lambda>(hops, dip, dsn, oip, sip). Rrep hops dip dsn oip sip"
+
+lemma rrep_simp [simp]:
+  "rrep(hops, dip, dsn, oip, sip) = Rrep hops dip dsn oip sip"
+  unfolding rrep_def by simp
+
+definition rerr :: "(ip \<rightharpoonup> sqn) \<times> ip \<Rightarrow> msg"
+  where "rerr \<equiv> \<lambda>(dests, sip). Rerr dests sip"
+
+lemma rerr_simp [simp]:
+  "rerr(dests, sip) = Rerr dests sip"
+  unfolding rerr_def by simp
+
+lemma not_eq_newpkt_rreq [simp]: "\<not>eq_newpkt (Rreq hops rreqid dip dsn dsk oip osn sip)"
+  unfolding eq_newpkt_def by simp
+
+lemma not_eq_newpkt_rrep [simp]: "\<not>eq_newpkt (Rrep hops dip dsn oip sip)"
+  unfolding eq_newpkt_def by simp
+
+lemma not_eq_newpkt_rerr [simp]: "\<not>eq_newpkt (Rerr dests sip)"
+  unfolding eq_newpkt_def by simp
+
+lemma not_eq_newpkt_pkt [simp]: "\<not>eq_newpkt (Pkt d dip sip)"
+  unfolding eq_newpkt_def by simp
+
+definition pkt :: "data \<times> ip \<times> ip \<Rightarrow> msg"
+  where "pkt \<equiv> \<lambda>(d, dip, sip). Pkt d dip sip"
+
+lemma pkt_simp [simp]:
+  "pkt(d, dip, sip) = Pkt d dip sip"
+  unfolding pkt_def by simp
+
+end

formal/afp/AODV/Aodv_Predicates.thy

@@ -0,0 +1,136 @@
+(*  Title:       Aodv_Predicates.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "Invariant assumptions and properties"
+
+theory Aodv_Predicates
+imports Aodv
+begin
+
+text \<open>Definitions for expression assumptions on incoming messages and properties of
+      outgoing messages.\<close>
+
+abbreviation not_Pkt :: "msg \<Rightarrow> bool"
+where "not_Pkt m \<equiv> case m of Pkt _ _ _ \<Rightarrow> False | _ \<Rightarrow> True"
+
+definition msg_sender :: "msg \<Rightarrow> ip"
+where "msg_sender m \<equiv> case m of Rreq _ _ _ _ _ _ _ ipc \<Rightarrow> ipc
+                              | Rrep _ _ _ _ ipc \<Rightarrow> ipc
+                              | Rerr _ ipc \<Rightarrow> ipc
+                              | Pkt _ _ ipc \<Rightarrow> ipc"
+
+lemma msg_sender_simps [simp]:
+  "\<And>hops rreqid dip dsn dsk oip osn sip.
+                          msg_sender (Rreq hops rreqid dip dsn dsk oip osn sip) = sip"
+  "\<And>hops dip dsn oip sip. msg_sender (Rrep hops dip dsn oip sip) = sip"
+  "\<And>dests sip.            msg_sender (Rerr dests sip) = sip"
+  "\<And>d dip sip.            msg_sender (Pkt d dip sip) = sip"
+  unfolding msg_sender_def by simp_all
+
+definition msg_zhops :: "msg \<Rightarrow> bool"
+where "msg_zhops m \<equiv> case m of
+                                 Rreq hopsc _ dipc _ _ oipc _ sipc \<Rightarrow> hopsc = 0 \<longrightarrow> oipc = sipc
+                               | Rrep hopsc dipc _ _ sipc \<Rightarrow> hopsc = 0 \<longrightarrow> dipc = sipc
+                               | _ \<Rightarrow> True"
+
+lemma msg_zhops_simps [simp]:
+  "\<And>hops rreqid dip dsn dsk oip osn sip.
+           msg_zhops (Rreq hops rreqid dip dsn dsk oip osn sip) = (hops = 0 \<longrightarrow> oip = sip)"
+  "\<And>hops dip dsn oip sip. msg_zhops (Rrep hops dip dsn oip sip) = (hops = 0 \<longrightarrow> dip = sip)"
+  "\<And>dests sip.            msg_zhops (Rerr dests sip)        = True"
+  "\<And>d dip.                msg_zhops (Newpkt d dip)          = True"
+  "\<And>d dip sip.            msg_zhops (Pkt d dip sip)         = True"
+  unfolding msg_zhops_def by simp_all
+
+definition rreq_rrep_sn :: "msg \<Rightarrow> bool"
+where "rreq_rrep_sn m \<equiv> case m of Rreq _ _ _ _ _ _ osnc _ \<Rightarrow> osnc \<ge> 1
+                                | Rrep _ _ dsnc _ _ \<Rightarrow> dsnc \<ge> 1
+                                | _ \<Rightarrow> True"
+
+lemma rreq_rrep_sn_simps [simp]:
+  "\<And>hops rreqid dip dsn dsk oip osn sip.
+           rreq_rrep_sn (Rreq hops rreqid dip dsn dsk oip osn sip) = (osn \<ge> 1)"
+  "\<And>hops dip dsn oip sip. rreq_rrep_sn (Rrep hops dip dsn oip sip) = (dsn \<ge> 1)"
+  "\<And>dests sip.            rreq_rrep_sn (Rerr dests sip) = True"
+  "\<And>d dip.                rreq_rrep_sn (Newpkt d dip)   = True"
+  "\<And>d dip sip.            rreq_rrep_sn (Pkt d dip sip)  = True"
+  unfolding rreq_rrep_sn_def by simp_all
+
+definition rreq_rrep_fresh :: "rt \<Rightarrow> msg \<Rightarrow> bool"
+where "rreq_rrep_fresh crt m \<equiv> case m of Rreq hopsc _ _ _ _ oipc osnc ipcc \<Rightarrow> (ipcc \<noteq> oipc \<longrightarrow>
+                                                oipc\<in>kD(crt) \<and> (sqn crt oipc > osnc
+                                                                \<or> (sqn crt oipc = osnc
+                                                                   \<and> the (dhops crt oipc) \<le> hopsc
+                                                                   \<and> the (flag crt oipc) = val)))
+                                | Rrep hopsc dipc dsnc _ ipcc \<Rightarrow> (ipcc \<noteq> dipc \<longrightarrow> 
+                                                                    dipc\<in>kD(crt)
+                                                                  \<and> sqn crt dipc = dsnc
+                                                                  \<and> the (dhops crt dipc) = hopsc
+                                                                  \<and> the (flag crt dipc) = val)
+                                | _ \<Rightarrow> True"
+
+lemma rreq_rrep_fresh [simp]:
+  "\<And>hops rreqid dip dsn dsk oip osn sip.
+           rreq_rrep_fresh crt (Rreq hops rreqid dip dsn dsk oip osn sip) =
+                               (sip \<noteq> oip \<longrightarrow> oip\<in>kD(crt)
+                                            \<and> (sqn crt oip > osn
+                                               \<or> (sqn crt oip = osn
+                                                  \<and> the (dhops crt oip) \<le> hops
+                                                  \<and> the (flag crt oip) = val)))"
+  "\<And>hops dip dsn oip sip. rreq_rrep_fresh crt (Rrep hops dip dsn oip sip) =
+                               (sip \<noteq> dip \<longrightarrow> dip\<in>kD(crt)
+                                              \<and> sqn crt dip = dsn
+                                              \<and> the (dhops crt dip) = hops
+                                              \<and> the (flag crt dip) = val)"
+  "\<And>dests sip.            rreq_rrep_fresh crt (Rerr dests sip) = True"
+  "\<And>d dip.                rreq_rrep_fresh crt (Newpkt d dip)   = True"
+  "\<And>d dip sip.            rreq_rrep_fresh crt (Pkt d dip sip)  = True"
+  unfolding rreq_rrep_fresh_def by simp_all
+
+definition rerr_invalid :: "rt \<Rightarrow> msg \<Rightarrow> bool"
+where "rerr_invalid crt m \<equiv> case m of Rerr destsc _ \<Rightarrow> (\<forall>ripc\<in>dom(destsc).
+                                            (ripc\<in>iD(crt) \<and> the (destsc ripc) = sqn crt ripc))
+                                | _ \<Rightarrow> True"
+
+lemma rerr_invalid [simp]:
+  "\<And>hops rreqid dip dsn dsk oip osn sip.
+                           rerr_invalid crt (Rreq hops rreqid dip dsn dsk oip osn sip) = True"
+  "\<And>hops dip dsn oip sip. rerr_invalid crt (Rrep hops dip dsn oip sip) = True"
+  "\<And>dests sip.            rerr_invalid crt (Rerr dests sip) = (\<forall>rip\<in>dom(dests).
+                                                 rip\<in>iD(crt) \<and> the (dests rip) = sqn crt rip)"
+  "\<And>d dip.                rerr_invalid crt (Newpkt d dip)   = True"
+  "\<And>d dip sip.            rerr_invalid crt (Pkt d dip sip)  = True"
+  unfolding rerr_invalid_def by simp_all
+
+definition
+  initmissing :: "(nat \<Rightarrow> state option) \<times> 'a \<Rightarrow> (nat \<Rightarrow> state) \<times> 'a"
+where
+  "initmissing \<sigma> = (\<lambda>i. case (fst \<sigma>) i of None \<Rightarrow> aodv_init i | Some s \<Rightarrow> s, snd \<sigma>)"
+
+lemma not_in_net_ips_fst_init_missing [simp]:
+  assumes "i \<notin> net_ips \<sigma>"
+    shows "fst (initmissing (netgmap fst \<sigma>)) i = aodv_init i"
+  using assms unfolding initmissing_def by simp
+
+lemma fst_initmissing_netgmap_pair_fst [simp]:
+  "fst (initmissing (netgmap (\<lambda>(p, q). (fst (id p), snd (id p), q)) s))
+                                               = fst (initmissing (netgmap fst s))"
+  unfolding initmissing_def by auto
+
+text \<open>We introduce a streamlined alternative to @{term initmissing} with @{term netgmap}
+        to simplify invariant statements and thus facilitate their comprehension and
+        presentation.\<close>
+
+lemma fst_initmissing_netgmap_default_aodv_init_netlift:
+  "fst (initmissing (netgmap fst s)) = default aodv_init (netlift fst s)"
+  unfolding initmissing_def default_def
+  by (simp add: fst_netgmap_netlift del: One_nat_def)
+
+definition
+  netglobal :: "((nat \<Rightarrow> state) \<Rightarrow> bool) \<Rightarrow> ((state \<times> 'b) \<times> 'c) net_state \<Rightarrow> bool"
+where
+  "netglobal P \<equiv> (\<lambda>s. P (default aodv_init (netlift fst s)))"
+
+end

formal/afp/AODV/Fresher.thy

@@ -0,0 +1,798 @@
+(*  Title:       Fresher.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "Quality relations between routes"
+
+theory Fresher
+imports Aodv_Data
+begin
+
+subsection "Net sequence numbers"
+
+subsubsection "On individual routes"
+
+definition
+  nsqn\<^sub>r :: "r \<Rightarrow> sqn"
+where
+  "nsqn\<^sub>r r \<equiv> if \<pi>\<^sub>4(r) = val \<or> \<pi>\<^sub>2(r) = 0 then \<pi>\<^sub>2(r) else (\<pi>\<^sub>2(r) - 1)"
+
+lemma nsqnr_def':
+  "nsqn\<^sub>r r = (if \<pi>\<^sub>4(r) = inv then \<pi>\<^sub>2(r) - 1 else \<pi>\<^sub>2(r))"
+  unfolding nsqn\<^sub>r_def by simp
+
+lemma nsqn\<^sub>r_zero [simp]:
+  "\<And>dsn dsk flag hops nhip pre. nsqn\<^sub>r (0, dsk, flag, hops, nhip, pre) = 0"
+  unfolding nsqn\<^sub>r_def by clarsimp
+
+lemma nsqn\<^sub>r_val [simp]:
+  "\<And>dsn dsk hops nhip pre. nsqn\<^sub>r (dsn, dsk, val, hops, nhip, pre) = dsn"
+  unfolding nsqn\<^sub>r_def by clarsimp
+
+lemma nsqn\<^sub>r_inv [simp]:
+  "\<And>dsn dsk hops nhip pre. nsqn\<^sub>r (dsn, dsk, inv, hops, nhip, pre) = dsn - 1"
+  unfolding nsqn\<^sub>r_def by clarsimp
+
+lemma nsqn\<^sub>r_lte_dsn [simp]:
+  "\<And>dsn dsk flag hops nhip pre. nsqn\<^sub>r (dsn, dsk, flag, hops, nhip, pre) \<le> dsn"
+  unfolding nsqn\<^sub>r_def by clarsimp
+
+subsubsection "On routes in routing tables"
+
+definition
+  nsqn :: "rt \<Rightarrow> ip \<Rightarrow> sqn"
+where
+  "nsqn \<equiv> \<lambda>rt dip. case \<sigma>\<^bsub>route\<^esub>(rt, dip) of None \<Rightarrow> 0 | Some r \<Rightarrow> nsqn\<^sub>r(r)"
+
+lemma nsqn_sqn_def:
+  "\<And>rt dip. nsqn rt dip = (if flag rt dip = Some val \<or> sqn rt dip = 0
+                            then sqn rt dip else sqn rt dip - 1)"
+  unfolding nsqn_def sqn_def by (clarsimp split: option.split)
+
+lemma not_in_kD_nsqn [simp]:
+  assumes "dip \<notin> kD(rt)"
+  shows "nsqn rt dip = 0"
+  using assms unfolding nsqn_def by simp
+
+lemma kD_nsqn:
+  assumes "dip \<in> kD(rt)"
+    shows "nsqn rt dip = nsqn\<^sub>r(the (\<sigma>\<^bsub>route\<^esub>(rt, dip)))"
+  using assms [THEN kD_Some] unfolding nsqn_def by clarsimp
+
+lemma nsqnr_r_flag_pred [simp, intro]:
+    fixes dsn dsk flag hops nhip pre
+  assumes "P (nsqn\<^sub>r (dsn, dsk, val, hops, nhip, pre))"
+      and "P (nsqn\<^sub>r (dsn, dsk, inv, hops, nhip, pre))"
+    shows "P (nsqn\<^sub>r (dsn, dsk, flag, hops, nhip, pre))"
+  using assms by (cases flag) auto
+
+lemma nsqn\<^sub>r_addpreRT_inv [simp]:
+  "\<And>rt dip npre dip'. dip \<in> kD(rt) \<Longrightarrow>
+   nsqn\<^sub>r (the (the (addpreRT rt dip npre) dip')) = nsqn\<^sub>r (the (rt dip'))"
+  unfolding addpreRT_def nsqn\<^sub>r_def
+  by (frule kD_Some) (clarsimp split: option.split)
+
+lemma sqn_nsqn:
+  "\<And>rt dip. sqn rt dip - 1 \<le> nsqn rt dip"
+  unfolding sqn_def nsqn_def by (clarsimp split: option.split)
+
+lemma nsqn_sqn: "nsqn rt dip \<le> sqn rt dip"
+  unfolding sqn_def nsqn_def by (cases "rt dip") auto
+
+lemma val_nsqn_sqn [elim, simp]:
+  assumes "ip\<in>kD(rt)"
+      and "the (flag rt ip) = val"
+    shows "nsqn rt ip = sqn rt ip"
+  using assms unfolding nsqn_sqn_def by auto
+
+lemma vD_nsqn_sqn [elim, simp]:
+  assumes "ip\<in>vD(rt)"
+    shows "nsqn rt ip = sqn rt ip"
+  proof -
+    from \<open>ip\<in>vD(rt)\<close> have "ip\<in>kD(rt)"
+                      and "the (flag rt ip) = val" by auto
+    thus ?thesis ..
+  qed
+
+lemma inv_nsqn_sqn [elim, simp]:
+  assumes "ip\<in>kD(rt)"
+      and "the (flag rt ip) = inv"
+    shows "nsqn rt ip = sqn rt ip - 1"
+  using assms unfolding nsqn_sqn_def by auto
+
+lemma iD_nsqn_sqn [elim, simp]:
+  assumes "ip\<in>iD(rt)"
+    shows "nsqn rt ip = sqn rt ip - 1"
+  proof -
+    from \<open>ip\<in>iD(rt)\<close> have "ip\<in>kD(rt)"
+                      and "the (flag rt ip) = inv" by auto
+    thus ?thesis ..
+  qed
+
+lemma nsqn_update_changed_kno_val [simp]: "\<And>rt ip dsn dsk hops nhip.
+  rt \<noteq> update rt ip (dsn, kno, val, hops, nhip, {})
+   \<Longrightarrow> nsqn (update rt ip (dsn, kno, val, hops, nhip, {})) ip = dsn"
+  unfolding nsqn\<^sub>r_def update_def
+  by (clarsimp simp: kD_nsqn split: option.split_asm option.split if_split_asm)
+     (metis fun_upd_triv)
+
+lemma nsqn_addpreRT_inv [simp]:
+  "\<And>rt dip npre dip'. dip \<in> kD(rt) \<Longrightarrow>
+   nsqn (the (addpreRT rt dip npre)) dip' = nsqn rt dip'"
+  unfolding addpreRT_def nsqn_def nsqn\<^sub>r_def
+  by (frule kD_Some) (clarsimp split: option.split)
+
+lemma nsqn_update_other [simp]:
+    fixes dsn dsk flag hops dip nhip pre rt ip
+  assumes "dip \<noteq> ip"
+    shows "nsqn (update rt ip (dsn, dsk, flag, hops, nhip, pre)) dip = nsqn rt dip"
+    using assms unfolding nsqn_def
+    by (clarsimp split: option.split)
+
+lemma nsqn_invalidate_eq:
+  assumes "dip \<in> kD(rt)"
+      and "dests dip = Some rsn"
+    shows "nsqn (invalidate rt dests) dip = rsn - 1"
+  using assms
+  proof -
+    from assms obtain dsk hops nhip pre
+      where "invalidate rt dests dip = Some (rsn, dsk, inv, hops, nhip, pre)"
+      unfolding invalidate_def by auto
+    moreover from \<open>dip \<in> kD(rt)\<close> have "dip \<in> kD(invalidate rt dests)" by simp
+    ultimately show ?thesis
+      using \<open>dests dip = Some rsn\<close> by simp
+  qed
+
+lemma nsqn_invalidate_other [simp]:
+  assumes "dip\<in>kD(rt)"
+      and "dip\<notin>dom dests"
+    shows "nsqn (invalidate rt dests) dip = nsqn rt dip"
+  using assms by (clarsimp simp add: kD_nsqn)
+
+subsection "Comparing routes "
+
+definition
+  fresher :: "r \<Rightarrow> r \<Rightarrow> bool" ("(_/ \<sqsubseteq> _)"  [51, 51] 50)
+where
+  "fresher r r' \<equiv> ((nsqn\<^sub>r r < nsqn\<^sub>r r') \<or> (nsqn\<^sub>r r  = nsqn\<^sub>r r' \<and> \<pi>\<^sub>5(r) \<ge> \<pi>\<^sub>5(r')))"
+
+lemma fresherI1 [intro]:
+  assumes "nsqn\<^sub>r r < nsqn\<^sub>r r'"
+    shows "r \<sqsubseteq> r'"
+  unfolding fresher_def using assms by simp
+
+lemma fresherI2 [intro]:
+  assumes "nsqn\<^sub>r r = nsqn\<^sub>r r'"
+      and "\<pi>\<^sub>5(r) \<ge> \<pi>\<^sub>5(r')"
+    shows "r \<sqsubseteq> r'"
+  unfolding fresher_def using assms by simp
+
+lemma fresherI [intro]:
+  assumes "(nsqn\<^sub>r r < nsqn\<^sub>r r') \<or> (nsqn\<^sub>r r  = nsqn\<^sub>r r' \<and> \<pi>\<^sub>5(r) \<ge> \<pi>\<^sub>5(r'))"
+    shows "r \<sqsubseteq> r'"
+  unfolding fresher_def using assms .
+
+lemma fresherE [elim]:
+  assumes "r \<sqsubseteq> r'"
+      and "nsqn\<^sub>r r < nsqn\<^sub>r r' \<Longrightarrow> P r r'"
+      and "nsqn\<^sub>r r  = nsqn\<^sub>r r' \<and> \<pi>\<^sub>5(r) \<ge> \<pi>\<^sub>5(r') \<Longrightarrow> P r r'"
+    shows "P r r'"
+  using assms unfolding fresher_def by auto
+
+lemma fresher_refl [simp]: "r \<sqsubseteq> r"
+  unfolding fresher_def by simp
+
+lemma fresher_trans [elim, trans]:
+  "\<lbrakk> x \<sqsubseteq> y; y \<sqsubseteq> z \<rbrakk> \<Longrightarrow> x \<sqsubseteq> z"
+  unfolding fresher_def by auto
+
+lemma not_fresher_trans [elim, trans]:
+  "\<lbrakk> \<not>(x \<sqsubseteq> y); \<not>(z \<sqsubseteq> x) \<rbrakk> \<Longrightarrow> \<not>(z \<sqsubseteq> y)"
+  unfolding fresher_def by auto
+
+lemma fresher_dsn_flag_hops_const [simp]:
+  fixes dsn dsk dsk' flag hops nhip nhip' pre pre'
+  shows "(dsn, dsk, flag, hops, nhip, pre) \<sqsubseteq> (dsn, dsk', flag, hops, nhip', pre')"
+  unfolding fresher_def by (cases flag) simp_all
+
+lemma addpre_fresher [simp]: "\<And>r npre. r \<sqsubseteq> (addpre r npre)"
+  by clarsimp
+
+subsection "Comparing routing tables "
+
+definition
+  rt_fresher :: "ip \<Rightarrow> rt \<Rightarrow> rt \<Rightarrow> bool"
+where
+  "rt_fresher \<equiv> \<lambda>dip rt rt'. (the (\<sigma>\<^bsub>route\<^esub>(rt, dip))) \<sqsubseteq> (the (\<sigma>\<^bsub>route\<^esub>(rt', dip)))"
+
+abbreviation
+   rt_fresher_syn :: "rt \<Rightarrow> ip \<Rightarrow> rt \<Rightarrow> bool" ("(_/ \<sqsubseteq>\<^bsub>_\<^esub> _)"  [51, 999, 51] 50)
+where
+  "rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2 \<equiv> rt_fresher i rt1 rt2"
+
+lemma rt_fresher_def':
+  "(rt\<^sub>1 \<sqsubseteq>\<^bsub>i\<^esub> rt\<^sub>2) = (nsqn\<^sub>r (the (rt\<^sub>1 i)) < nsqn\<^sub>r (the (rt\<^sub>2 i)) \<or>
+                     nsqn\<^sub>r (the (rt\<^sub>1 i)) = nsqn\<^sub>r (the (rt\<^sub>2 i)) \<and> \<pi>\<^sub>5 (the (rt\<^sub>2 i)) \<le> \<pi>\<^sub>5 (the (rt\<^sub>1 i)))"
+  unfolding rt_fresher_def fresher_def by (rule refl)
+
+lemma single_rt_fresher [intro]:
+  assumes "the (rt1 ip) \<sqsubseteq> the (rt2 ip)"
+    shows "rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2"
+  using assms unfolding rt_fresher_def .
+
+lemma rt_fresher_single [intro]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2"
+    shows "the (rt1 ip) \<sqsubseteq> the (rt2 ip)"
+  using assms unfolding rt_fresher_def .
+
+lemma rt_fresher_def2:
+  assumes "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+    shows "(rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2) = (nsqn rt1 dip < nsqn rt2 dip
+                               \<or> (nsqn rt1 dip = nsqn rt2 dip
+                                   \<and> the (dhops rt1 dip) \<ge> the (dhops rt2 dip)))"
+  using assms unfolding rt_fresher_def fresher_def by (simp add: kD_nsqn proj5_eq_dhops)
+
+lemma rt_fresherI1 [intro]:
+  assumes "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+      and "nsqn rt1 dip < nsqn rt2 dip"
+    shows "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+  unfolding rt_fresher_def2 [OF assms(1-2)] using assms(3) by simp
+
+lemma rt_fresherI2 [intro]:
+  assumes "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+      and "nsqn rt1 dip = nsqn rt2 dip"
+      and "the (dhops rt1 dip) \<ge> the (dhops rt2 dip)"
+    shows "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+  unfolding rt_fresher_def2 [OF assms(1-2)] using assms(3-4) by simp
+
+lemma rt_fresherE [elim]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+      and "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+      and "\<lbrakk> nsqn rt1 dip < nsqn rt2 dip \<rbrakk> \<Longrightarrow> P rt1 rt2 dip"
+      and "\<lbrakk> nsqn rt1 dip = nsqn rt2 dip;
+             the (dhops rt1 dip) \<ge> the (dhops rt2 dip) \<rbrakk> \<Longrightarrow> P rt1 rt2 dip"
+    shows "P rt1 rt2 dip"
+  using assms(1) unfolding rt_fresher_def2 [OF assms(2-3)]
+  using assms(4-5) by auto
+
+lemma rt_fresher_refl [simp]: "rt \<sqsubseteq>\<^bsub>dip\<^esub> rt"
+  unfolding rt_fresher_def by simp
+
+lemma rt_fresher_trans [elim, trans]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+      and "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt3"
+    shows "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt3"
+  using assms unfolding rt_fresher_def by auto
+
+lemma rt_fresher_if_Some [intro!]:
+  assumes "the (rt dip) \<sqsubseteq> r"
+    shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> (\<lambda>ip. if ip = dip then Some r else rt ip)"
+  using assms unfolding rt_fresher_def by simp
+
+definition rt_fresh_as :: "ip \<Rightarrow> rt \<Rightarrow> rt \<Rightarrow> bool"
+where
+  "rt_fresh_as \<equiv> \<lambda>dip rt1 rt2. (rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2) \<and> (rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)"
+
+abbreviation
+   rt_fresh_as_syn :: "rt \<Rightarrow> ip \<Rightarrow> rt \<Rightarrow> bool" ("(_/ \<approx>\<^bsub>_\<^esub> _)"  [51, 999, 51] 50)
+where
+  "rt1 \<approx>\<^bsub>i\<^esub> rt2 \<equiv> rt_fresh_as i rt1 rt2"
+
+lemma rt_fresh_as_refl [simp]: "\<And>rt dip. rt \<approx>\<^bsub>dip\<^esub> rt"
+  unfolding rt_fresh_as_def by simp
+
+lemma rt_fresh_as_trans [simp, intro, trans]:
+  "\<And>rt1 rt2 rt3 dip. \<lbrakk> rt1 \<approx>\<^bsub>dip\<^esub> rt2; rt2 \<approx>\<^bsub>dip\<^esub> rt3 \<rbrakk> \<Longrightarrow> rt1 \<approx>\<^bsub>dip\<^esub> rt3"
+  unfolding rt_fresh_as_def rt_fresher_def
+  by (metis (mono_tags) fresher_trans)
+
+lemma rt_fresh_asI [intro!]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+      and "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1"
+    shows "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+  using assms unfolding rt_fresh_as_def by simp
+
+lemma rt_fresh_as_fresherI [intro]:
+  assumes "dip\<in>kD(rt1)"
+      and "dip\<in>kD(rt2)"
+      and "the (rt1 dip) \<sqsubseteq> the (rt2 dip)"
+      and "the (rt2 dip) \<sqsubseteq> the (rt1 dip)"
+    shows "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+  using assms unfolding rt_fresh_as_def
+  by (clarsimp dest!: single_rt_fresher)
+
+lemma nsqn_rt_fresh_asI:
+  assumes "dip \<in> kD(rt)"
+      and "dip \<in> kD(rt')"
+      and "nsqn rt dip = nsqn rt' dip"
+      and "\<pi>\<^sub>5(the (rt dip)) = \<pi>\<^sub>5(the (rt' dip))"
+    shows "rt \<approx>\<^bsub>dip\<^esub> rt'"
+  proof
+    from assms(1-2,4) have dhops': "the (dhops rt' dip) \<le> the (dhops rt dip)"
+      by (simp add: proj5_eq_dhops)
+    with assms(1-3) show "rt \<sqsubseteq>\<^bsub>dip\<^esub> rt'"
+      by (rule rt_fresherI2)
+  next
+    from assms(1-2,4) have dhops: "the (dhops rt dip) \<le> the (dhops rt' dip)"
+      by (simp add: proj5_eq_dhops)
+    with assms(2,1) assms(3) [symmetric] show "rt' \<sqsubseteq>\<^bsub>dip\<^esub> rt"
+      by (rule rt_fresherI2)
+  qed
+
+lemma rt_fresh_asE [elim]:
+  assumes "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+      and "\<lbrakk> rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2; rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1 \<rbrakk> \<Longrightarrow> P rt1 rt2 dip"
+    shows "P rt1 rt2 dip"
+  using assms unfolding rt_fresh_as_def by simp
+
+lemma rt_fresh_asD1 [dest]:
+  assumes "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    shows "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+  using assms unfolding rt_fresh_as_def by simp
+
+lemma rt_fresh_asD2 [dest]:
+  assumes "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    shows "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1"
+  using assms unfolding rt_fresh_as_def by simp
+
+lemma rt_fresh_as_sym:
+  assumes "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    shows "rt2 \<approx>\<^bsub>dip\<^esub> rt1"
+  using assms unfolding rt_fresh_as_def by simp
+
+lemma not_rt_fresh_asI1 [intro]:
+  assumes "\<not> (rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2)"
+    shows "\<not> (rt1 \<approx>\<^bsub>dip\<^esub> rt2)"
+  proof
+    assume "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    hence "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2" ..
+    with \<open>\<not> (rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2)\<close> show False ..
+  qed
+
+lemma not_rt_fresh_asI2 [intro]:
+  assumes "\<not> (rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)"
+    shows "\<not> (rt1 \<approx>\<^bsub>dip\<^esub> rt2)"
+  proof
+    assume "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    hence "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1" ..
+    with \<open>\<not> (rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)\<close> show False ..
+  qed
+
+lemma not_single_rt_fresher [elim]:
+  assumes "\<not>(the (rt1 ip) \<sqsubseteq> the (rt2 ip))"
+    shows "\<not>(rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2)"
+  proof
+    assume "rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2"
+    hence "the (rt1 ip) \<sqsubseteq> the (rt2 ip)" ..
+    with \<open>\<not>(the (rt1 ip) \<sqsubseteq> the (rt2 ip))\<close> show False ..
+  qed
+
+lemmas not_single_rt_fresh_asI1 [intro] =  not_rt_fresh_asI1 [OF not_single_rt_fresher]
+lemmas not_single_rt_fresh_asI2 [intro] =  not_rt_fresh_asI2 [OF not_single_rt_fresher]
+
+lemma not_rt_fresher_single [elim]:
+  assumes "\<not>(rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2)"
+    shows "\<not>(the (rt1 ip) \<sqsubseteq> the (rt2 ip))"
+  proof
+    assume "the (rt1 ip) \<sqsubseteq> the (rt2 ip)"
+    hence "rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2" ..
+    with \<open>\<not>(rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2)\<close> show False ..
+  qed
+
+lemma rt_fresh_as_nsqnr:
+  assumes "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+      and "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    shows "nsqn\<^sub>r (the (rt2 dip)) = nsqn\<^sub>r (the (rt1 dip))"
+  using assms(3) unfolding rt_fresh_as_def
+  by (auto simp: rt_fresher_def2 [OF \<open>dip \<in> kD(rt1)\<close> \<open>dip \<in> kD(rt2)\<close>]
+                 rt_fresher_def2 [OF \<open>dip \<in> kD(rt2)\<close> \<open>dip \<in> kD(rt1)\<close>]
+                 kD_nsqn [OF \<open>dip \<in> kD(rt1)\<close>]
+                 kD_nsqn [OF \<open>dip \<in> kD(rt2)\<close>])
+
+lemma rt_fresher_mapupd [intro!]:
+  assumes "dip\<in>kD(rt)"
+      and "the (rt dip) \<sqsubseteq> r"
+    shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> rt(dip \<mapsto> r)"
+  using assms unfolding rt_fresher_def by simp
+
+lemma rt_fresher_map_update_other [intro!]:
+  assumes "dip\<in>kD(rt)"
+      and "dip \<noteq> ip"
+    shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> rt(ip \<mapsto> r)"
+  using assms unfolding rt_fresher_def by simp
+
+lemma rt_fresher_update_other [simp]:
+  assumes inkD: "dip\<in>kD(rt)"
+     and "dip \<noteq> ip"
+   shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> update rt ip r"
+   using assms unfolding update_def
+   by (clarsimp split: option.split) (fastforce)
+
+theorem rt_fresher_update [simp]:
+  assumes "dip\<in>kD(rt)"
+      and "the (dhops rt dip) \<ge> 1"
+      and "update_arg_wf r"
+   shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> update rt ip r"
+  proof (cases "dip = ip")
+    assume "dip \<noteq> ip" with \<open>dip\<in>kD(rt)\<close> show ?thesis
+      by (rule rt_fresher_update_other)
+  next
+    assume "dip = ip"
+
+    from \<open>dip\<in>kD(rt)\<close> obtain dsn\<^sub>n dsk\<^sub>n f\<^sub>n hops\<^sub>n nhip\<^sub>n pre\<^sub>n
+      where rtn [simp]: "the (rt dip) = (dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)"
+      by (metis prod_cases6)
+    with \<open>the (dhops rt dip) \<ge> 1\<close> and \<open>dip\<in>kD(rt)\<close> have "hops\<^sub>n \<ge> 1"
+      by (metis proj5_eq_dhops projs(4))
+    from \<open>dip\<in>kD(rt)\<close> rtn have [simp]: "sqn rt dip = dsn\<^sub>n"
+                           and [simp]: "the (dhops rt dip) = hops\<^sub>n"
+                           and [simp]: "the (flag rt dip) = f\<^sub>n"
+      by (simp add: sqn_def proj5_eq_dhops [symmetric]
+                            proj4_eq_flag [symmetric])+
+
+    from \<open>update_arg_wf r\<close> have "(dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+                                  \<sqsubseteq> the ((update rt dip r) dip)"
+      proof (rule wf_r_cases)
+        fix nhip pre
+        from \<open>hops\<^sub>n \<ge> 1\<close> have "\<And>pre'. (dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+                                        \<sqsubseteq> (dsn\<^sub>n, unk, val, Suc 0, nhip, pre')"
+          unfolding fresher_def sqn_def by (cases f\<^sub>n) auto
+        thus "(dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+               \<sqsubseteq> the (update rt dip (0, unk, val, Suc 0, nhip, pre) dip)"
+          using \<open>dip\<in>kD(rt)\<close> by - (rule update_cases_kD, simp_all)
+      next
+        fix dsn :: sqn and hops nhip pre
+        assume "0 < dsn"
+        show "(dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+               \<sqsubseteq> the (update rt dip (dsn, kno, val, hops, nhip, pre) dip)"
+        proof (rule update_cases_kD [OF _ \<open>dip\<in>kD(rt)\<close>], simp_all add: \<open>0 < dsn\<close>)
+          assume "dsn\<^sub>n < dsn"
+            thus "(dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+                   \<sqsubseteq> (dsn, kno, val, hops, nhip, pre \<union> pre\<^sub>n)"
+              unfolding fresher_def by auto
+        next
+          assume "dsn\<^sub>n = dsn"
+             and "hops < hops\<^sub>n"
+            thus "(dsn, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+                   \<sqsubseteq> (dsn, kno, val, hops, nhip, pre \<union> pre\<^sub>n)"
+              unfolding fresher_def nsqn\<^sub>r_def by simp
+        next
+          assume "dsn\<^sub>n = dsn"
+            with \<open>0 < dsn\<close>
+            show "(dsn, dsk\<^sub>n, inv, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+                   \<sqsubseteq> (dsn, kno, val, hops, nhip, pre \<union> pre\<^sub>n)"
+              unfolding fresher_def by simp
+        qed
+      qed
+    hence "rt \<sqsubseteq>\<^bsub>dip\<^esub> update rt dip r"
+      by - (rule single_rt_fresher, simp)
+    with \<open>dip = ip\<close> show ?thesis by simp
+  qed
+
+theorem rt_fresher_invalidate [simp]:
+  assumes "dip\<in>kD(rt)"
+      and indests: "\<forall>rip\<in>dom(dests). rip\<in>vD(rt) \<and> sqn rt rip < the (dests rip)"
+    shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> invalidate rt dests"
+  proof (cases "dip\<in>dom(dests)")
+    assume "dip\<notin>dom(dests)"
+      thus ?thesis using \<open>dip\<in>kD(rt)\<close>
+      by - (rule single_rt_fresher, simp)
+  next
+    assume "dip\<in>dom(dests)"
+    moreover with indests have "dip\<in>vD(rt)"
+                           and "sqn rt dip < the (dests dip)"
+      by auto
+    ultimately show ?thesis
+      unfolding invalidate_def sqn_def
+      by - (rule single_rt_fresher, auto simp: fresher_def)
+  qed
+
+lemma nsqn\<^sub>r_invalidate [simp]:
+  assumes "dip\<in>kD(rt)"
+      and "dip\<in>dom(dests)"
+    shows "nsqn\<^sub>r (the (invalidate rt dests dip)) = the (dests dip) - 1"
+  using assms unfolding invalidate_def by auto
+
+lemma rt_fresh_as_inc_invalidate [simp]:
+  assumes "dip\<in>kD(rt)"
+      and "\<forall>rip\<in>dom(dests). rip\<in>vD(rt) \<and> the (dests rip) = inc (sqn rt rip)"
+    shows "rt \<approx>\<^bsub>dip\<^esub> invalidate rt dests"
+  proof (cases "dip\<in>dom(dests)")
+    assume "dip\<notin>dom(dests)"
+    with \<open>dip\<in>kD(rt)\<close> have "dip\<in>kD(invalidate rt dests)"
+      by simp
+    with \<open>dip\<in>kD(rt)\<close> show ?thesis
+      by rule (simp_all add: \<open>dip\<notin>dom(dests)\<close>)
+  next
+    assume "dip\<in>dom(dests)"
+    with assms(2) have "dip\<in>vD(rt)"
+                  and "the (dests dip) = inc (sqn rt dip)" by auto
+    from \<open>dip\<in>vD(rt)\<close> have "dip\<in>kD(rt)" by simp
+    moreover then have "dip\<in>kD(invalidate rt dests)" by simp
+    ultimately show ?thesis
+    proof (rule nsqn_rt_fresh_asI)
+      from \<open>dip\<in>vD(rt)\<close> have "nsqn rt dip = sqn rt dip" by simp
+      also have "sqn rt dip = nsqn\<^sub>r (the (invalidate rt dests dip))"
+      proof -
+        from \<open>dip\<in>kD(rt)\<close> have "nsqn\<^sub>r (the (invalidate rt dests dip)) = the (dests dip) - 1"
+          using \<open>dip\<in>dom(dests)\<close> by (rule nsqn\<^sub>r_invalidate)
+        with \<open>the (dests dip) = inc (sqn rt dip)\<close>
+          show "sqn rt dip = nsqn\<^sub>r (the (invalidate rt dests dip))" by simp
+      qed
+      also from \<open>dip\<in>kD(invalidate rt dests)\<close>
+        have "nsqn\<^sub>r (the (invalidate rt dests dip)) = nsqn (invalidate rt dests) dip"
+          by (simp add: kD_nsqn)
+      finally show "nsqn rt dip = nsqn (invalidate rt dests) dip" .
+    qed simp
+  qed
+
+lemmas rt_fresher_inc_invalidate [simp] = rt_fresh_as_inc_invalidate [THEN rt_fresh_asD1]
+
+lemma rt_fresh_as_addpreRT [simp]:
+  assumes "ip\<in>kD(rt)"
+    shows "rt \<approx>\<^bsub>dip\<^esub> the (addpreRT rt ip npre)"
+  using assms [THEN kD_Some] by (auto simp: addpreRT_def)
+
+lemmas rt_fresher_addpreRT [simp] = rt_fresh_as_addpreRT [THEN rt_fresh_asD1]
+
+subsection "Strictly comparing routing tables "
+
+definition rt_strictly_fresher :: "ip \<Rightarrow> rt \<Rightarrow> rt \<Rightarrow> bool"
+where
+  "rt_strictly_fresher \<equiv> \<lambda>dip rt1 rt2. (rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2) \<and> \<not>(rt1 \<approx>\<^bsub>dip\<^esub> rt2)"
+
+abbreviation
+   rt_strictly_fresher_syn :: "rt \<Rightarrow> ip \<Rightarrow> rt \<Rightarrow> bool" ("(_/ \<sqsubset>\<^bsub>_\<^esub> _)"  [51, 999, 51] 50)
+where
+  "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2 \<equiv> rt_strictly_fresher i rt1 rt2"
+
+lemma rt_strictly_fresher_def'':
+  "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2 = ((rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2) \<and> \<not>(rt2 \<sqsubseteq>\<^bsub>i\<^esub> rt1))"
+  unfolding rt_strictly_fresher_def rt_fresh_as_def by auto
+
+lemma rt_strictly_fresherI' [intro]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2"
+      and "\<not>(rt2 \<sqsubseteq>\<^bsub>i\<^esub> rt1)"
+    shows "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2"
+  using assms unfolding rt_strictly_fresher_def'' by simp
+
+lemma rt_strictly_fresherE' [elim]:
+  assumes "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2"
+      and "\<lbrakk> rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2; \<not>(rt2 \<sqsubseteq>\<^bsub>i\<^esub> rt1) \<rbrakk> \<Longrightarrow> P rt1 rt2 i"
+    shows "P rt1 rt2 i"
+  using assms unfolding rt_strictly_fresher_def'' by simp
+
+lemma rt_strictly_fresherI [intro]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2"
+      and "\<not>(rt1 \<approx>\<^bsub>i\<^esub> rt2)"
+    shows "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2"
+  unfolding rt_strictly_fresher_def using assms ..
+
+lemmas rt_strictly_fresher_singleI [elim] = rt_strictly_fresherI [OF single_rt_fresher]
+
+lemma rt_strictly_fresherE [elim]:
+  assumes "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2"
+      and "\<lbrakk> rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2; \<not>(rt1 \<approx>\<^bsub>i\<^esub> rt2) \<rbrakk> \<Longrightarrow> P rt1 rt2 i"
+    shows "P rt1 rt2 i"
+  using assms(1) unfolding rt_strictly_fresher_def
+  by rule (erule(1) assms(2))
+
+lemma rt_strictly_fresher_def':
+  "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2 =
+     (nsqn\<^sub>r (the (rt1 i)) < nsqn\<^sub>r (the (rt2 i))
+       \<or> (nsqn\<^sub>r (the (rt1 i)) = nsqn\<^sub>r (the (rt2 i)) \<and> \<pi>\<^sub>5(the (rt1 i)) > \<pi>\<^sub>5(the (rt2 i))))"
+  unfolding rt_strictly_fresher_def'' rt_fresher_def fresher_def by auto
+
+lemma rt_strictly_fresher_fresherD [dest]:
+  assumes "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2"
+    shows "the (rt1 dip) \<sqsubseteq> the (rt2 dip)"
+  using assms unfolding rt_strictly_fresher_def rt_fresher_def by auto
+
+lemma rt_strictly_fresher_not_fresh_asD [dest]:
+  assumes "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2"
+    shows "\<not> rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+  using assms unfolding rt_strictly_fresher_def by auto
+
+lemma rt_strictly_fresher_trans [elim, trans]:
+  assumes "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2"
+      and "rt2 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+    shows "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+  using assms proof -
+    from \<open>rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2\<close> obtain "the (rt1 dip) \<sqsubseteq> the (rt2 dip)" by auto
+    also from \<open>rt2 \<sqsubset>\<^bsub>dip\<^esub> rt3\<close> obtain "the (rt2 dip) \<sqsubseteq> the (rt3 dip)" by auto
+    finally have "the (rt1 dip) \<sqsubseteq> the (rt3 dip)" .
+
+    moreover have "\<not> (rt1 \<approx>\<^bsub>dip\<^esub> rt3)"
+    proof -    
+      from \<open>rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2\<close> obtain "\<not>(the (rt2 dip) \<sqsubseteq> the (rt1 dip))" by auto
+      also from \<open>rt2 \<sqsubset>\<^bsub>dip\<^esub> rt3\<close> obtain "\<not>(the (rt3 dip) \<sqsubseteq> the (rt2 dip))" by auto
+      finally have "\<not>(the (rt3 dip) \<sqsubseteq> the (rt1 dip))" .
+      thus ?thesis ..
+    qed
+    ultimately show "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3" ..
+ qed
+
+lemma rt_strictly_fresher_irefl [simp]: "\<not> (rt \<sqsubset>\<^bsub>dip\<^esub> rt)"
+  unfolding rt_strictly_fresher_def
+  by clarsimp
+
+lemma rt_fresher_trans_rt_strictly_fresher [elim, trans]:
+  assumes "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2"
+      and "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt3"
+    shows "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+  proof -
+    from \<open>rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2\<close> have "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+                           and "\<not>(rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)"
+      unfolding rt_strictly_fresher_def'' by auto
+    from this(1) and \<open>rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt3\<close> have "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt3" ..
+
+    moreover from \<open>\<not>(rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)\<close> have "\<not>(rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)"
+      proof (rule contrapos_nn)
+        assume "rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt1"
+        with \<open>rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt3\<close> show "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1" ..
+      qed
+
+    ultimately show "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+      unfolding rt_strictly_fresher_def'' by auto
+  qed
+
+lemma rt_fresher_trans_rt_strictly_fresher' [elim, trans]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+      and "rt2 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+    shows "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+  proof -
+    from \<open>rt2 \<sqsubset>\<^bsub>dip\<^esub> rt3\<close> have "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt3"
+                           and "\<not>(rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt2)"
+      unfolding rt_strictly_fresher_def'' by auto
+    from \<open>rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2\<close> and this(1) have "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt3" ..
+
+    moreover from \<open>\<not>(rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt2)\<close> have "\<not>(rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)"
+      proof (rule contrapos_nn)
+        assume "rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt1"
+        thus "rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt2" using \<open>rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2\<close> ..
+      qed
+
+    ultimately show "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+      unfolding rt_strictly_fresher_def'' by auto
+  qed
+
+lemma rt_fresher_imp_nsqn_le:
+  assumes "rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2"
+      and "ip \<in> kD rt1"
+      and "ip \<in> kD rt2"
+    shows "nsqn rt1 ip \<le> nsqn rt2 ip"
+  using assms(1)
+  by (auto simp add: rt_fresher_def2 [OF assms(2-3)])
+
+lemma rt_strictly_fresher_ltI [intro]:
+  assumes "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+      and "nsqn rt1 dip < nsqn rt2 dip"
+    shows "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2"
+  proof
+    from assms show "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2" ..
+  next
+    show "\<not>(rt1 \<approx>\<^bsub>dip\<^esub> rt2)"
+    proof
+      assume "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+      hence "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1" ..
+      hence "nsqn rt2 dip \<le> nsqn rt1 dip"
+        using \<open>dip \<in> kD(rt2)\<close> \<open>dip \<in> kD(rt1)\<close>
+        by (rule rt_fresher_imp_nsqn_le)
+      with \<open>nsqn rt1 dip < nsqn rt2 dip\<close> show "False"
+        by simp
+    qed
+  qed
+
+lemma rt_strictly_fresher_eqI [intro]:
+  assumes "i\<in>kD(rt1)"
+      and "i\<in>kD(rt2)"
+      and "nsqn rt1 i = nsqn rt2 i"
+      and "\<pi>\<^sub>5(the (rt2 i)) < \<pi>\<^sub>5(the (rt1 i))"
+    shows "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2"
+  using assms unfolding rt_strictly_fresher_def' by (auto simp add: kD_nsqn)
+
+lemma invalidate_rtsf_left [simp]:
+  "\<And>dests dip rt rt'. dests dip = None \<Longrightarrow> (invalidate rt dests \<sqsubset>\<^bsub>dip\<^esub> rt') = (rt \<sqsubset>\<^bsub>dip\<^esub> rt')"
+  unfolding invalidate_def rt_strictly_fresher_def'
+  by (rule iffI) (auto split: option.split_asm)
+
+lemma vD_invalidate_rt_strictly_fresher [simp]:
+  assumes "dip \<in> vD(invalidate rt1 dests)"
+    shows "(invalidate rt1 dests \<sqsubset>\<^bsub>dip\<^esub> rt2) = (rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2)"
+  proof (cases "dip \<in> dom(dests)")
+    assume "dip \<in> dom(dests)"
+    hence "dip \<notin> vD(invalidate rt1 dests)"
+      unfolding invalidate_def vD_def
+      by clarsimp (metis assms option.simps(3) vD_invalidate_vD_not_dests)
+    with \<open>dip \<in> vD(invalidate rt1 dests)\<close> show ?thesis by simp
+  next
+    assume "dip \<notin> dom(dests)"
+    hence "dests dip = None" by auto
+    moreover with \<open>dip \<in> vD(invalidate rt1 dests)\<close> have "dip \<in> vD(rt1)"
+      unfolding invalidate_def vD_def
+      by clarsimp (metis (opaque_lifting, no_types) assms vD_Some vD_invalidate_vD_not_dests)
+    ultimately show ?thesis
+      unfolding invalidate_def rt_strictly_fresher_def' by auto
+  qed
+
+lemma rt_strictly_fresher_update_other [elim!]:
+  "\<And>dip ip rt r rt'. \<lbrakk> dip \<noteq> ip; rt \<sqsubset>\<^bsub>dip\<^esub> rt' \<rbrakk> \<Longrightarrow> update rt ip r \<sqsubset>\<^bsub>dip\<^esub> rt'"
+  unfolding rt_strictly_fresher_def' by clarsimp
+
+lemma addpreRT_strictly_fresher [simp]:
+  assumes "dip \<in> kD(rt)"
+    shows "(the (addpreRT rt dip npre) \<sqsubset>\<^bsub>ip\<^esub> rt2) = (rt \<sqsubset>\<^bsub>ip\<^esub> rt2)"
+  using assms unfolding rt_strictly_fresher_def' by clarsimp
+
+lemma lt_sqn_imp_update_strictly_fresher:
+  assumes "dip \<in> vD (rt2 nhip)"
+      and  *: "osn < sqn (rt2 nhip) dip"
+      and **: "rt \<noteq> update rt dip (osn, kno, val, hops, nhip, {})"
+    shows "update rt dip (osn, kno, val, hops, nhip, {}) \<sqsubset>\<^bsub>dip\<^esub> rt2 nhip"
+  unfolding rt_strictly_fresher_def'
+  proof (rule disjI1)
+    from ** have "nsqn (update rt dip (osn, kno, val, hops, nhip, {})) dip = osn"
+      by (rule nsqn_update_changed_kno_val)
+    with \<open>dip\<in>vD(rt2 nhip)\<close>
+      have "nsqn\<^sub>r (the (update rt dip (osn, kno, val, hops, nhip, {}) dip)) = osn"
+        by (simp add: kD_nsqn)
+    also have "osn < sqn (rt2 nhip) dip" by (rule *)
+    also have "sqn (rt2 nhip) dip = nsqn\<^sub>r (the (rt2 nhip dip))"
+      unfolding nsqn\<^sub>r_def using \<open>dip \<in> vD (rt2 nhip)\<close>
+      by - (metis vD_flag_val proj2_eq_sqn proj4_eq_flag vD_iD_gives_kD(1))
+    finally show "nsqn\<^sub>r (the (update rt dip (osn, kno, val, hops, nhip, {}) dip))
+                                                              < nsqn\<^sub>r (the (rt2 nhip dip))" .
+  qed
+
+lemma dhops_le_hops_imp_update_strictly_fresher:
+  assumes "dip \<in> vD(rt2 nhip)"
+      and sqn: "sqn (rt2 nhip) dip = osn"
+      and hop: "the (dhops (rt2 nhip) dip) \<le> hops"
+      and **: "rt \<noteq> update rt dip (osn, kno, val, Suc hops, nhip, {})"
+    shows "update rt dip (osn, kno, val, Suc hops, nhip, {}) \<sqsubset>\<^bsub>dip\<^esub> rt2 nhip"
+  unfolding rt_strictly_fresher_def'
+  proof (rule disjI2, rule conjI)
+    from ** have "nsqn (update rt dip (osn, kno, val, Suc hops, nhip, {})) dip = osn"
+      by (rule nsqn_update_changed_kno_val)
+    with \<open>dip\<in>vD(rt2 nhip)\<close>
+      have "nsqn\<^sub>r (the (update rt dip (osn, kno, val, Suc hops, nhip, {}) dip)) = osn"
+        by (simp add: kD_nsqn)
+    also have "osn = sqn (rt2 nhip) dip" by (rule sqn [symmetric])
+    also have "sqn (rt2 nhip) dip = nsqn\<^sub>r (the (rt2 nhip dip))"
+      unfolding nsqn\<^sub>r_def using \<open>dip \<in> vD(rt2 nhip)\<close>
+      by - (metis vD_flag_val proj2_eq_sqn proj4_eq_flag vD_iD_gives_kD(1))
+    finally show "nsqn\<^sub>r (the (update rt dip (osn, kno, val, Suc hops, nhip, {}) dip))
+                                                              = nsqn\<^sub>r (the (rt2 nhip dip))" .
+  next
+    have "the (dhops (rt2 nhip) dip) \<le> hops" by (rule hop)
+    also have "hops < hops + 1" by simp
+    also have "hops + 1 = the (dhops (update rt dip (osn, kno, val, Suc hops, nhip, {})) dip)"
+      using ** by simp
+    finally have "the (dhops (rt2 nhip) dip)
+                        < the (dhops (update rt dip (osn, kno, val, Suc hops, nhip, {})) dip)" .
+    thus "\<pi>\<^sub>5 (the (rt2 nhip dip)) < \<pi>\<^sub>5 (the (update rt dip (osn, kno, val, Suc hops, nhip, {}) dip))"
+      using \<open>dip \<in> vD(rt2 nhip)\<close> by (simp add: proj5_eq_dhops)
+  qed
+
+lemma nsqn_invalidate:
+  assumes "dip \<in> kD(rt)"
+      and "\<forall>ip\<in>dom(dests). ip \<in> vD(rt) \<and> the (dests ip) = inc (sqn rt ip)"
+    shows "nsqn (invalidate rt dests) dip = nsqn rt dip"
+  proof -
+    from \<open>dip \<in> kD(rt)\<close> have "dip \<in> kD(invalidate rt dests)" by simp
+
+    from assms have "rt \<approx>\<^bsub>dip\<^esub> invalidate rt dests"
+      by (rule rt_fresh_as_inc_invalidate)
+    with \<open>dip \<in> kD(rt)\<close> \<open>dip \<in> kD(invalidate rt dests)\<close> show ?thesis
+      by (simp add: kD_nsqn del: invalidate_kD_inv)
+         (erule(2) rt_fresh_as_nsqnr)
+  qed
+
+end

formal/afp/AODV/Global_Invariants.thy

@@ -0,0 +1,1151 @@
+(*  Title:       Global_Invariants.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "Global invariant proofs over sequential processes"
+
+theory Global_Invariants
+imports Seq_Invariants
+        Aodv_Predicates
+        Fresher
+        Quality_Increases
+        AWN.OAWN_Convert
+        OAodv
+begin
+
+lemma other_quality_increases [elim]:
+  assumes "other quality_increases I \<sigma> \<sigma>'"
+    shows "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+  using assms by (rule, clarsimp) (metis quality_increases_refl)
+
+lemma weaken_otherwith [elim]:
+    fixes m
+  assumes *: "otherwith P I (orecvmsg Q) \<sigma> \<sigma>' a"
+      and weakenP: "\<And>\<sigma> m. P \<sigma> m \<Longrightarrow> P' \<sigma> m"
+      and weakenQ: "\<And>\<sigma> m. Q \<sigma> m \<Longrightarrow> Q' \<sigma> m"
+    shows "otherwith P' I (orecvmsg Q') \<sigma> \<sigma>' a"
+  proof
+    fix j
+    assume "j\<notin>I"
+    with * have "P (\<sigma> j) (\<sigma>' j)" by auto
+    thus "P' (\<sigma> j) (\<sigma>' j)" by (rule weakenP)
+  next
+    from * have "orecvmsg Q \<sigma> a" by auto
+    thus "orecvmsg Q' \<sigma> a"
+      by rule (erule weakenQ)
+  qed
+
+lemma oreceived_msg_inv:
+  assumes other: "\<And>\<sigma> \<sigma>' m. \<lbrakk> P \<sigma> m; other Q {i} \<sigma> \<sigma>' \<rbrakk> \<Longrightarrow> P \<sigma>' m"
+      and local: "\<And>\<sigma> m. P \<sigma> m \<Longrightarrow> P (\<sigma>(i := \<sigma> i\<lparr>msg := m\<rparr>)) m"
+    shows "opaodv i \<Turnstile> (otherwith Q {i} (orecvmsg P), other Q {i} \<rightarrow>)
+                       onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<sigma>, l). l \<in> {PAodv-:1} \<longrightarrow> P \<sigma> (msg (\<sigma> i)))"
+  proof (inv_cterms, intro impI)
+    fix \<sigma> \<sigma>' l
+    assume "l = PAodv-:1 \<longrightarrow> P \<sigma> (msg (\<sigma> i))"
+       and "l = PAodv-:1"
+       and "other Q {i} \<sigma> \<sigma>'"
+    from this(1-2) have "P \<sigma> (msg (\<sigma> i))" ..
+    hence "P \<sigma>' (msg (\<sigma> i))" using \<open>other Q {i} \<sigma> \<sigma>'\<close>
+      by (rule other)
+    moreover from \<open>other Q {i} \<sigma> \<sigma>'\<close> have "\<sigma>' i = \<sigma> i" ..
+    ultimately show "P \<sigma>' (msg (\<sigma>' i))" by simp
+  next
+    fix \<sigma> \<sigma>' msg
+    assume "otherwith Q {i} (orecvmsg P) \<sigma> \<sigma>' (receive msg)"
+       and "\<sigma>' i = \<sigma> i\<lparr>msg := msg\<rparr>"
+    from this(1) have "P \<sigma> msg"
+                 and "\<forall>j. j\<noteq>i \<longrightarrow> Q (\<sigma> j) (\<sigma>' j)" by auto
+    from this(1) have "P (\<sigma>(i := \<sigma> i\<lparr>msg := msg\<rparr>)) msg" by (rule local)
+    thus "P \<sigma>' msg"
+    proof (rule other)
+      from \<open>\<sigma>' i = \<sigma> i\<lparr>msg := msg\<rparr>\<close> and \<open>\<forall>j. j\<noteq>i \<longrightarrow> Q (\<sigma> j) (\<sigma>' j)\<close>
+        show "other Q {i} (\<sigma>(i := \<sigma> i\<lparr>msg := msg\<rparr>)) \<sigma>'"
+          by - (rule otherI, auto)
+    qed
+  qed
+
+
+text \<open>(Equivalent to) Proposition 7.27\<close>
+
+lemma local_quality_increases:
+  "paodv i \<TTurnstile>\<^sub>A (recvmsg rreq_rrep_sn \<rightarrow>) onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). quality_increases \<xi> \<xi>')"
+  proof (rule step_invariantI)
+    fix s a s'
+    assume sr: "s \<in> reachable (paodv i) (recvmsg rreq_rrep_sn)"
+       and tr: "(s, a, s') \<in> trans (paodv i)"
+       and rm: "recvmsg rreq_rrep_sn a"
+    from sr have srTT: "s \<in> reachable (paodv i) TT" ..
+
+    from route_tables_fresher sr tr rm
+      have "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). \<forall>dip\<in>kD (rt \<xi>). rt \<xi> \<sqsubseteq>\<^bsub>dip\<^esub> rt \<xi>') (s, a, s')"
+        by (rule step_invariantD)
+
+    moreover from known_destinations_increase srTT tr TT_True
+      have "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). kD (rt \<xi>) \<subseteq> kD (rt \<xi>')) (s, a, s')"
+        by (rule step_invariantD)
+
+    moreover from sqns_increase srTT tr TT_True
+      have "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). \<forall>ip. sqn (rt \<xi>) ip \<le> sqn (rt \<xi>') ip) (s, a, s')"
+        by (rule step_invariantD)
+
+     ultimately show "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). quality_increases \<xi> \<xi>') (s, a, s')"
+       unfolding onll_def by auto
+  qed
+
+lemmas olocal_quality_increases =
+   open_seq_step_invariant [OF local_quality_increases initiali_aodv oaodv_trans aodv_trans,
+                            simplified seqll_onll_swap]
+
+lemma oquality_increases:
+  "opaodv i \<Turnstile>\<^sub>A (otherwith quality_increases {i} (orecvmsg (\<lambda>_. rreq_rrep_sn)),
+                other quality_increases {i} \<rightarrow>)
+                onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<sigma>, _), _, (\<sigma>', _)). \<forall>j. quality_increases (\<sigma> j) (\<sigma>' j))"
+  (is "_ \<Turnstile>\<^sub>A (?S, _ \<rightarrow>) _")
+  proof (rule onll_ostep_invariantI, simp)
+    fix \<sigma> p l a \<sigma>' p' l'
+    assume or: "(\<sigma>, p) \<in> oreachable (opaodv i) ?S (other quality_increases {i})"
+       and ll: "l \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p"
+       and "?S \<sigma> \<sigma>' a"
+       and tr: "((\<sigma>, p), a, (\<sigma>', p')) \<in> oseqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i"
+       and ll': "l' \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p'"
+    from this(1-3) have "orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma> a"
+      by (auto dest!: oreachable_weakenE [where QS="act (recvmsg rreq_rrep_sn)"
+                                            and QU="other quality_increases {i}"]
+                      otherwith_actionD)
+    with or have orw: "(\<sigma>, p) \<in> oreachable (opaodv i) (act (recvmsg rreq_rrep_sn))
+                                                      (other quality_increases {i})"
+      by - (erule oreachable_weakenE, auto)
+    with tr ll ll' and \<open>orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma> a\<close> have "quality_increases (\<sigma> i) (\<sigma>' i)"
+      by - (drule onll_ostep_invariantD [OF olocal_quality_increases], auto simp: seqll_def)
+    with \<open>?S \<sigma> \<sigma>' a\<close> show "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+      by (auto dest!: otherwith_syncD)
+  qed
+
+lemma rreq_rrep_nsqn_fresh_any_step_invariant:
+  "opaodv i \<Turnstile>\<^sub>A (act (recvmsg rreq_rrep_sn), other A {i} \<rightarrow>)
+                onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<sigma>, _), a, _). anycast (msg_fresh \<sigma>) a)"
+  proof (rule ostep_invariantI, simp del: act_simp)
+    fix \<sigma> p a \<sigma>' p'
+    assume or: "(\<sigma>, p) \<in> oreachable (opaodv i) (act (recvmsg rreq_rrep_sn)) (other A {i})"
+       and "((\<sigma>, p), a, (\<sigma>', p')) \<in> oseqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i"
+       and recv: "act (recvmsg rreq_rrep_sn) \<sigma> \<sigma>' a"
+    obtain l l' where "l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p" and "l'\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p'"
+      by (metis aodv_ex_label)
+    from \<open>((\<sigma>, p), a, (\<sigma>', p')) \<in> oseqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i\<close>
+      have tr: "((\<sigma>, p), a, (\<sigma>', p')) \<in> trans (opaodv i)" by simp
+
+    have "anycast (rreq_rrep_fresh (rt (\<sigma> i))) a"
+    proof -
+      have "opaodv i \<Turnstile>\<^sub>A (act (recvmsg rreq_rrep_sn), other A {i} \<rightarrow>)
+                           onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (seqll i (\<lambda>((\<xi>, _), a, _). anycast (rreq_rrep_fresh (rt \<xi>)) a))"
+        by (rule ostep_invariant_weakenE [OF
+              open_seq_step_invariant [OF rreq_rrep_fresh_any_step_invariant initiali_aodv,
+                                       simplified seqll_onll_swap]]) auto
+      hence "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (seqll i (\<lambda>((\<xi>, _), a, _). anycast (rreq_rrep_fresh (rt \<xi>)) a))
+                                                                     ((\<sigma>, p), a, (\<sigma>', p'))"
+        using or tr recv by - (erule(4) ostep_invariantE)
+      thus ?thesis
+        using \<open>l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p\<close> and \<open>l'\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p'\<close> by auto
+    qed
+
+    moreover have "anycast (rerr_invalid (rt (\<sigma> i))) a"
+    proof -
+      have "opaodv i \<Turnstile>\<^sub>A (act (recvmsg rreq_rrep_sn), other A {i} \<rightarrow>)
+                           onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (seqll i (\<lambda>((\<xi>, _), a, _). anycast (rerr_invalid (rt \<xi>)) a))"
+        by (rule ostep_invariant_weakenE [OF
+              open_seq_step_invariant [OF rerr_invalid_any_step_invariant initiali_aodv,
+                                       simplified seqll_onll_swap]]) auto
+      hence "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (seqll i (\<lambda>((\<xi>, _), a, _). anycast (rerr_invalid (rt \<xi>)) a))
+                                                                     ((\<sigma>, p), a, (\<sigma>', p'))"
+        using or tr recv by - (erule(4) ostep_invariantE)
+      thus ?thesis
+        using \<open>l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p\<close> and \<open>l'\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p'\<close> by auto
+    qed
+
+    moreover have "anycast rreq_rrep_sn a"
+    proof -
+      from or tr recv
+        have "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (seqll i (\<lambda>(_, a, _). anycast rreq_rrep_sn a)) ((\<sigma>, p), a, (\<sigma>', p'))"
+          by (rule ostep_invariantE [OF
+                open_seq_step_invariant [OF rreq_rrep_sn_any_step_invariant initiali_aodv
+                                            oaodv_trans aodv_trans,
+                                         simplified seqll_onll_swap]])
+      thus ?thesis
+        using \<open>l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p\<close> and \<open>l'\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p'\<close> by auto
+    qed
+
+    moreover have "anycast (\<lambda>m. not_Pkt m \<longrightarrow> msg_sender m = i) a"
+      proof -
+        have "opaodv i \<Turnstile>\<^sub>A (act (recvmsg rreq_rrep_sn), other A {i} \<rightarrow>)
+               onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (seqll i (\<lambda>((\<xi>, _), a, _). anycast (\<lambda>m. not_Pkt m \<longrightarrow> msg_sender m = i) a))"
+          by (rule ostep_invariant_weakenE [OF
+                open_seq_step_invariant [OF sender_ip_valid initiali_aodv,
+                                         simplified seqll_onll_swap]]) auto
+        thus ?thesis using or tr recv \<open>l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p\<close> and \<open>l'\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p'\<close>
+          by - (drule(3) onll_ostep_invariantD, auto)
+      qed
+
+    ultimately have "anycast (msg_fresh \<sigma>) a"
+      by (simp_all add: anycast_def
+                   del: msg_fresh
+                   split: seq_action.split_asm msg.split_asm) simp_all
+    thus "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<sigma>, _), a, _). anycast (msg_fresh \<sigma>) a) ((\<sigma>, p), a, (\<sigma>', p'))"
+      by auto
+  qed
+
+lemma oreceived_rreq_rrep_nsqn_fresh_inv:
+  "opaodv i \<Turnstile> (otherwith quality_increases {i} (orecvmsg msg_fresh),
+                other quality_increases {i} \<rightarrow>)
+               onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<sigma>, l). l \<in> {PAodv-:1} \<longrightarrow> msg_fresh \<sigma> (msg (\<sigma> i)))"
+  proof (rule oreceived_msg_inv)
+    fix \<sigma> \<sigma>' m
+    assume *: "msg_fresh \<sigma> m"
+       and "other quality_increases {i} \<sigma> \<sigma>'"
+    from this(2) have "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)" ..
+    thus "msg_fresh \<sigma>' m" using * ..
+  next
+    fix \<sigma> m
+    assume "msg_fresh \<sigma> m"
+    thus "msg_fresh (\<sigma>(i := \<sigma> i\<lparr>msg := m\<rparr>)) m"
+    proof (cases m)
+      fix dests sip
+      assume "m = Rerr dests sip"
+      with \<open>msg_fresh \<sigma> m\<close> show ?thesis
+        by auto
+    qed auto
+  qed
+
+lemma oquality_increases_nsqn_fresh:
+  "opaodv i \<Turnstile>\<^sub>A (otherwith quality_increases {i} (orecvmsg msg_fresh),
+                other quality_increases {i} \<rightarrow>)
+                onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<sigma>, _), _, (\<sigma>', _)). \<forall>j. quality_increases (\<sigma> j) (\<sigma>' j))"
+  by (rule ostep_invariant_weakenE [OF oquality_increases]) auto
+
+lemma oosn_rreq:
+  "opaodv i \<Turnstile> (otherwith quality_increases {i} (orecvmsg msg_fresh),
+                other quality_increases {i} \<rightarrow>)
+       onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (seql i (\<lambda>(\<xi>, l). l \<in> {PAodv-:4, PAodv-:5} \<union> {PRreq-:n |n. True} \<longrightarrow> 1 \<le> osn \<xi>))"
+  by (rule oinvariant_weakenE [OF open_seq_invariant [OF osn_rreq initiali_aodv]])
+     (auto simp: seql_onl_swap)
+
+lemma rreq_sip:
+  "opaodv i \<Turnstile> (otherwith quality_increases {i} (orecvmsg msg_fresh),
+                other quality_increases {i} \<rightarrow>)
+               onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<sigma>, l).
+                   (l \<in> {PAodv-:4, PAodv-:5, PRreq-:0, PRreq-:2} \<and> sip (\<sigma> i) \<noteq> oip (\<sigma> i))
+                    \<longrightarrow> oip (\<sigma> i) \<in> kD(rt (\<sigma> (sip (\<sigma> i))))
+                        \<and> nsqn (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma> i)) \<ge> osn (\<sigma> i)
+                        \<and> (nsqn (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma> i)) = osn (\<sigma> i)
+                            \<longrightarrow> (hops (\<sigma> i) \<ge> the (dhops (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma> i)))
+                                  \<or> the (flag (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma> i))) = inv)))"
+    (is "_ \<Turnstile> (?S, ?U \<rightarrow>) _")
+    proof (inv_cterms inv add: oseq_step_invariant_sterms [OF oquality_increases_nsqn_fresh
+                                                              aodv_wf oaodv_trans]
+                             onl_oinvariant_sterms [OF aodv_wf oreceived_rreq_rrep_nsqn_fresh_inv]
+                             onl_oinvariant_sterms [OF aodv_wf oosn_rreq]
+                   simp add: seqlsimp
+                   simp del: One_nat_def, rule impI)
+    fix \<sigma> \<sigma>' p l
+    assume "(\<sigma>, p) \<in> oreachable (opaodv i) ?S ?U"
+       and "l \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p"
+       and pre:
+           "(l = PAodv-:4 \<or> l = PAodv-:5 \<or> l = PRreq-:0 \<or> l = PRreq-:2) \<and> sip (\<sigma> i) \<noteq> oip (\<sigma> i)
+             \<longrightarrow> oip (\<sigma> i) \<in> kD (rt (\<sigma> (sip (\<sigma> i))))
+                 \<and> osn (\<sigma> i) \<le> nsqn (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma> i))
+                 \<and> (nsqn (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma> i)) = osn (\<sigma> i)
+                    \<longrightarrow> the (dhops (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma> i))) \<le> hops (\<sigma> i)
+                        \<or> the (flag (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma> i))) = inv)"
+       and "other quality_increases {i} \<sigma> \<sigma>'"
+       and hyp: "(l=PAodv-:4 \<or> l=PAodv-:5 \<or> l=PRreq-:0 \<or> l=PRreq-:2) \<and> sip (\<sigma>' i) \<noteq> oip (\<sigma>' i)"
+           (is "?labels \<and> sip (\<sigma>' i) \<noteq> oip (\<sigma>' i)")
+    from this(4) have "\<sigma>' i = \<sigma> i" ..
+    with hyp have hyp': "?labels \<and> sip (\<sigma> i) \<noteq> oip (\<sigma> i)" by simp
+    show "oip (\<sigma>' i) \<in> kD (rt (\<sigma>' (sip (\<sigma>' i))))
+          \<and> osn (\<sigma>' i) \<le> nsqn (rt (\<sigma>' (sip (\<sigma>' i)))) (oip (\<sigma>' i))
+          \<and> (nsqn (rt (\<sigma>' (sip (\<sigma>' i)))) (oip (\<sigma>' i)) = osn (\<sigma>' i)
+             \<longrightarrow> the (dhops (rt (\<sigma>' (sip (\<sigma>' i)))) (oip (\<sigma>' i))) \<le> hops (\<sigma>' i)
+                  \<or> the (flag (rt (\<sigma>' (sip (\<sigma>' i)))) (oip (\<sigma>' i))) = inv)"
+    proof (cases "sip (\<sigma> i) = i")
+      assume "sip (\<sigma> i) \<noteq> i"
+      from \<open>other quality_increases {i} \<sigma> \<sigma>'\<close>
+        have "quality_increases (\<sigma> (sip (\<sigma> i))) (\<sigma>' (sip (\<sigma>' i)))"
+          by (rule otherE)  (clarsimp simp: \<open>sip (\<sigma> i) \<noteq> i\<close>)
+      moreover from \<open>(\<sigma>, p) \<in> oreachable (opaodv i) ?S ?U\<close> \<open>l \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p\<close> and hyp
+        have "1 \<le> osn (\<sigma>' i)"
+          by (auto dest!: onl_oinvariant_weakenD [OF oosn_rreq]
+                   simp add: seqlsimp \<open>\<sigma>' i = \<sigma> i\<close>)
+      moreover from \<open>sip (\<sigma> i) \<noteq> i\<close> hyp' and pre
+        have "oip (\<sigma>' i) \<in> kD (rt (\<sigma> (sip (\<sigma> i))))
+              \<and> osn (\<sigma>' i) \<le> nsqn (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma>' i))
+              \<and> (nsqn (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma>' i)) = osn (\<sigma>' i)
+                  \<longrightarrow> the (dhops (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma>' i))) \<le> hops (\<sigma>' i)
+                       \<or> the (flag (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma>' i))) = inv)"
+        by (auto simp: \<open>\<sigma>' i = \<sigma> i\<close>)
+      ultimately show ?thesis
+        by (rule quality_increases_rreq_rrep_props)
+    next
+      assume "sip (\<sigma> i) = i" thus ?thesis
+        using \<open>\<sigma>' i = \<sigma> i\<close> hyp and pre by auto
+    qed
+  qed (auto elim!: quality_increases_rreq_rrep_props')
+
+lemma odsn_rrep:
+  "opaodv i \<Turnstile> (otherwith quality_increases {i} (orecvmsg msg_fresh),
+                other quality_increases {i} \<rightarrow>)
+      onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (seql i (\<lambda>(\<xi>, l). l \<in> {PAodv-:6, PAodv-:7} \<union> {PRrep-:n|n. True} \<longrightarrow> 1 \<le> dsn \<xi>))"
+  by (rule oinvariant_weakenE [OF open_seq_invariant [OF dsn_rrep initiali_aodv]])
+     (auto simp: seql_onl_swap)
+
+lemma rrep_sip:
+  "opaodv i \<Turnstile> (otherwith quality_increases {i} (orecvmsg msg_fresh),
+                other quality_increases {i} \<rightarrow>)
+               onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<sigma>, l).
+                   (l \<in> {PAodv-:6, PAodv-:7, PRrep-:0, PRrep-:1} \<and> sip (\<sigma> i) \<noteq> dip (\<sigma> i))
+                    \<longrightarrow> dip (\<sigma> i) \<in> kD(rt (\<sigma> (sip (\<sigma> i))))
+                        \<and> nsqn (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma> i)) \<ge> dsn (\<sigma> i)
+                        \<and> (nsqn (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma> i)) = dsn (\<sigma> i)
+                            \<longrightarrow> (hops (\<sigma> i) \<ge> the (dhops (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma> i)))
+                                  \<or> the (flag (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma> i))) = inv)))"
+    (is "_ \<Turnstile> (?S, ?U \<rightarrow>) _")
+  proof (inv_cterms inv add: oseq_step_invariant_sterms [OF oquality_increases_nsqn_fresh aodv_wf
+                                                            oaodv_trans]
+                             onl_oinvariant_sterms [OF aodv_wf oreceived_rreq_rrep_nsqn_fresh_inv]
+                             onl_oinvariant_sterms [OF aodv_wf odsn_rrep]
+                   simp del: One_nat_def, rule impI)
+    fix \<sigma> \<sigma>' p l
+    assume "(\<sigma>, p) \<in> oreachable (opaodv i) ?S ?U"
+       and "l \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p"
+       and pre:
+           "(l = PAodv-:6 \<or> l = PAodv-:7 \<or> l = PRrep-:0 \<or> l = PRrep-:1) \<and> sip (\<sigma> i) \<noteq> dip (\<sigma> i)
+           \<longrightarrow> dip (\<sigma> i) \<in> kD (rt (\<sigma> (sip (\<sigma> i))))
+               \<and> dsn (\<sigma> i) \<le> nsqn (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma> i))
+               \<and> (nsqn (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma> i)) = dsn (\<sigma> i)
+                  \<longrightarrow> the (dhops (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma> i))) \<le> hops (\<sigma> i)
+                      \<or> the (flag (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma> i))) = inv)"
+       and "other quality_increases {i} \<sigma> \<sigma>'"
+       and hyp: "(l=PAodv-:6 \<or> l=PAodv-:7 \<or> l=PRrep-:0 \<or> l=PRrep-:1) \<and> sip (\<sigma>' i) \<noteq> dip (\<sigma>' i)"
+           (is "?labels \<and> sip (\<sigma>' i) \<noteq> dip (\<sigma>' i)")
+    from this(4) have "\<sigma>' i = \<sigma> i" ..
+    with hyp have hyp': "?labels \<and> sip (\<sigma> i) \<noteq> dip (\<sigma> i)" by simp
+    show "dip (\<sigma>' i) \<in> kD (rt (\<sigma>' (sip (\<sigma>' i))))
+          \<and> dsn (\<sigma>' i) \<le> nsqn (rt (\<sigma>' (sip (\<sigma>' i)))) (dip (\<sigma>' i))
+          \<and> (nsqn (rt (\<sigma>' (sip (\<sigma>' i)))) (dip (\<sigma>' i)) = dsn (\<sigma>' i)
+             \<longrightarrow> the (dhops (rt (\<sigma>' (sip (\<sigma>' i)))) (dip (\<sigma>' i))) \<le> hops (\<sigma>' i)
+                 \<or> the (flag (rt (\<sigma>' (sip (\<sigma>' i)))) (dip (\<sigma>' i))) = inv)"
+    proof (cases "sip (\<sigma> i) = i")
+      assume "sip (\<sigma> i) \<noteq> i"
+      from \<open>other quality_increases {i} \<sigma> \<sigma>'\<close>
+        have "quality_increases (\<sigma> (sip (\<sigma> i))) (\<sigma>' (sip (\<sigma>' i)))"
+          by (rule otherE)  (clarsimp simp: \<open>sip (\<sigma> i) \<noteq> i\<close>)
+      moreover from \<open>(\<sigma>, p) \<in> oreachable (opaodv i) ?S ?U\<close> \<open>l \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p\<close> and hyp
+        have "1 \<le> dsn (\<sigma>' i)"
+          by (auto dest!: onl_oinvariant_weakenD [OF odsn_rrep]
+                   simp add: seqlsimp \<open>\<sigma>' i = \<sigma> i\<close>)
+      moreover from \<open>sip (\<sigma> i) \<noteq> i\<close> hyp' and pre
+        have "dip (\<sigma>' i) \<in> kD (rt (\<sigma> (sip (\<sigma> i))))
+              \<and> dsn (\<sigma>' i) \<le> nsqn (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma>' i))
+              \<and> (nsqn (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma>' i)) = dsn (\<sigma>' i)
+                 \<longrightarrow> the (dhops (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma>' i))) \<le> hops (\<sigma>' i)
+                     \<or> the (flag (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma>' i))) = inv)"
+        by (auto simp: \<open>\<sigma>' i = \<sigma> i\<close>)
+      ultimately show ?thesis
+        by (rule quality_increases_rreq_rrep_props)
+    next
+      assume "sip (\<sigma> i) = i" thus ?thesis
+        using \<open>\<sigma>' i = \<sigma> i\<close> hyp and pre by auto
+    qed
+  qed (auto simp add: seqlsimp elim!: quality_increases_rreq_rrep_props')
+
+lemma rerr_sip:
+  "opaodv i \<Turnstile> (otherwith quality_increases {i} (orecvmsg msg_fresh),
+                other quality_increases {i} \<rightarrow>)
+               onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<sigma>, l).
+                 l \<in> {PAodv-:8, PAodv-:9, PRerr-:0, PRerr-:1}
+                 \<longrightarrow> (\<forall>ripc\<in>dom(dests (\<sigma> i)). ripc\<in>kD(rt (\<sigma> (sip (\<sigma> i)))) \<and>
+                        the (dests (\<sigma> i) ripc) - 1 \<le> nsqn (rt (\<sigma> (sip (\<sigma> i)))) ripc))"
+  (is "_ \<Turnstile> (?S, ?U \<rightarrow>) _")
+  proof -
+    { fix dests rip sip rsn and \<sigma> \<sigma>' :: "ip \<Rightarrow> state"
+      assume qinc: "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+         and *: "\<forall>rip\<in>dom dests. rip \<in> kD (rt (\<sigma> sip))
+                                  \<and> the (dests rip) - 1 \<le> nsqn (rt (\<sigma> sip)) rip"
+         and "dests rip = Some rsn"
+      from this(3) have "rip\<in>dom dests" by auto
+      with * and \<open>dests rip = Some rsn\<close> have "rip\<in>kD(rt (\<sigma> sip))"
+                                         and "rsn - 1 \<le> nsqn (rt (\<sigma> sip)) rip"
+        by (auto dest!: bspec)
+      from qinc have "quality_increases (\<sigma> sip) (\<sigma>' sip)" ..
+      have "rip \<in> kD(rt (\<sigma>' sip)) \<and> rsn - 1 \<le> nsqn (rt (\<sigma>' sip)) rip"
+      proof
+        from \<open>rip\<in>kD(rt (\<sigma> sip))\<close> and \<open>quality_increases (\<sigma> sip) (\<sigma>' sip)\<close>
+          show "rip \<in> kD(rt (\<sigma>' sip))" ..
+      next
+        from \<open>rip\<in>kD(rt (\<sigma> sip))\<close> and \<open>quality_increases (\<sigma> sip) (\<sigma>' sip)\<close>
+          have "nsqn (rt (\<sigma> sip)) rip \<le> nsqn (rt (\<sigma>' sip)) rip" ..
+        with \<open>rsn - 1 \<le> nsqn (rt (\<sigma> sip)) rip\<close> show "rsn - 1 \<le> nsqn (rt (\<sigma>' sip)) rip"
+          by (rule le_trans)
+      qed
+    } note partial = this
+
+    show ?thesis
+      by (inv_cterms inv add: oseq_step_invariant_sterms [OF oquality_increases_nsqn_fresh aodv_wf
+                                                             oaodv_trans]
+                              onl_oinvariant_sterms [OF aodv_wf oreceived_rreq_rrep_nsqn_fresh_inv]
+                              other_quality_increases other_localD
+                    simp del: One_nat_def, intro conjI)
+         (clarsimp simp del: One_nat_def split: if_split_asm option.split_asm, erule(2) partial)+
+  qed
+
+lemma prerr_guard: "paodv i \<TTurnstile>
+                  onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). (l = PRerr-:1
+                      \<longrightarrow> (\<forall>ip\<in>dom(dests \<xi>). ip\<in>vD(rt \<xi>)
+                                             \<and> the (nhop (rt \<xi>) ip) = sip \<xi>
+                                             \<and> sqn (rt \<xi>) ip < the (dests \<xi> ip))))"
+  by (inv_cterms) (clarsimp split: option.split_asm if_split_asm)
+
+lemmas oaddpreRT_welldefined =
+         open_seq_invariant [OF addpreRT_welldefined initiali_aodv oaodv_trans aodv_trans,
+                             simplified seql_onl_swap,
+                             THEN oinvariant_anyact]
+
+lemmas odests_vD_inc_sqn =
+         open_seq_invariant [OF dests_vD_inc_sqn initiali_aodv oaodv_trans aodv_trans,
+                             simplified seql_onl_swap,
+                             THEN oinvariant_anyact]
+
+lemmas oprerr_guard =
+         open_seq_invariant [OF prerr_guard initiali_aodv oaodv_trans aodv_trans,
+                             simplified seql_onl_swap,
+                             THEN oinvariant_anyact]
+
+text \<open>Proposition 7.28\<close>
+
+lemma seq_compare_next_hop':
+  "opaodv i \<Turnstile> (otherwith quality_increases {i} (orecvmsg msg_fresh),
+                other quality_increases {i} \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<sigma>, _).
+                   \<forall>dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                         in dip \<in> kD(rt (\<sigma> i)) \<and> nhip \<noteq> dip \<longrightarrow>
+                            dip \<in> kD(rt (\<sigma> nhip)) \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> nhip)) dip)"
+  (is "_ \<Turnstile> (?S, ?U \<rightarrow>) _")
+  proof -
+
+  { fix nhop and \<sigma> \<sigma>' :: "ip \<Rightarrow> state"
+    assume  pre: "\<forall>dip\<in>kD(rt (\<sigma> i)). nhop dip \<noteq> dip \<longrightarrow>
+                    dip\<in>kD(rt (\<sigma> (nhop dip))) \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> (nhop dip))) dip"
+       and qinc: "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+    have "\<forall>dip\<in>kD(rt (\<sigma> i)). nhop dip \<noteq> dip \<longrightarrow>
+                  dip\<in>kD(rt (\<sigma>' (nhop dip))) \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip"
+    proof (intro ballI impI)
+      fix dip
+      assume "dip\<in>kD(rt (\<sigma> i))"
+         and "nhop dip \<noteq> dip"
+      with pre have "dip\<in>kD(rt (\<sigma> (nhop dip)))"
+                and "nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> (nhop dip))) dip"
+        by auto
+      from qinc have qinc_nhop: "quality_increases (\<sigma> (nhop dip)) (\<sigma>' (nhop dip))" ..
+      with \<open>dip\<in>kD(rt (\<sigma> (nhop dip)))\<close> have "dip\<in>kD (rt (\<sigma>' (nhop dip)))" ..
+
+      moreover have "nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip"
+      proof -
+        from \<open>dip\<in>kD(rt (\<sigma> (nhop dip)))\<close> qinc_nhop
+          have "nsqn (rt (\<sigma> (nhop dip))) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip" ..
+        with \<open>nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> (nhop dip))) dip\<close> show ?thesis
+          by simp
+      qed
+
+      ultimately show "dip\<in>kD(rt (\<sigma>' (nhop dip)))
+                       \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip" ..
+    qed
+  } note basic = this
+
+  { fix nhop and \<sigma> \<sigma>' :: "ip \<Rightarrow> state"
+    assume pre: "\<forall>dip\<in>kD(rt (\<sigma> i)). nhop dip \<noteq> dip \<longrightarrow> dip\<in>kD(rt (\<sigma> (nhop dip)))
+                                             \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> (nhop dip))) dip"
+       and ndest: "\<forall>ripc\<in>dom (dests (\<sigma> i)). ripc \<in> kD (rt (\<sigma> (sip (\<sigma> i))))
+                                   \<and> the (dests (\<sigma> i) ripc) - 1 \<le> nsqn (rt (\<sigma> (sip (\<sigma> i)))) ripc"
+       and issip: "\<forall>ip\<in>dom (dests (\<sigma> i)). nhop ip = sip (\<sigma> i)"
+       and qinc: "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+    have "\<forall>dip\<in>kD(rt (\<sigma> i)). nhop dip \<noteq> dip \<longrightarrow> dip \<in> kD (rt (\<sigma>' (nhop dip)))
+                 \<and> nsqn (invalidate (rt (\<sigma> i)) (dests (\<sigma> i))) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip"
+    proof (intro ballI impI)
+      fix dip
+      assume "dip\<in>kD(rt (\<sigma> i))"
+         and "nhop dip \<noteq> dip"
+      with pre and qinc have "dip\<in>kD(rt (\<sigma>' (nhop dip)))"
+                         and "nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip"
+        by (auto dest!: basic)
+
+      have "nsqn (invalidate (rt (\<sigma> i)) (dests (\<sigma> i))) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip"
+      proof (cases "dip\<in>dom (dests (\<sigma> i))")
+        assume "dip\<in>dom (dests (\<sigma> i))"
+        with \<open>dip\<in>kD(rt (\<sigma> i))\<close> obtain dsn where "dests (\<sigma> i) dip = Some dsn"
+          by auto
+        with \<open>dip\<in>kD(rt (\<sigma> i))\<close> have "nsqn (invalidate (rt (\<sigma> i)) (dests (\<sigma> i))) dip = dsn - 1"
+           by (rule nsqn_invalidate_eq)
+        moreover have "dsn - 1 \<le> nsqn (rt (\<sigma>' (nhop dip))) dip"
+        proof -
+          from \<open>dests (\<sigma> i) dip = Some dsn\<close> have "the (dests (\<sigma> i) dip) = dsn" by simp
+          with ndest and \<open>dip\<in>dom (dests (\<sigma> i))\<close> have "dip \<in> kD (rt (\<sigma> (sip (\<sigma> i))))"
+                                                      "dsn - 1 \<le> nsqn (rt (\<sigma> (sip (\<sigma> i)))) dip"
+            by auto
+          moreover from issip and \<open>dip\<in>dom (dests (\<sigma> i))\<close> have "nhop dip = sip (\<sigma> i)" ..
+          ultimately have "dip \<in> kD (rt (\<sigma> (nhop dip)))"
+                      and "dsn - 1 \<le> nsqn (rt (\<sigma> (nhop dip))) dip" by auto
+          with qinc show "dsn - 1 \<le> nsqn (rt (\<sigma>' (nhop dip))) dip"
+            by simp (metis kD_nsqn_quality_increases_trans)
+        qed
+        ultimately show ?thesis by simp
+      next
+        assume "dip \<notin> dom (dests (\<sigma> i))"
+        with \<open>dip\<in>kD(rt (\<sigma> i))\<close>
+          have "nsqn (invalidate (rt (\<sigma> i)) (dests (\<sigma> i))) dip = nsqn (rt (\<sigma> i)) dip"
+            by (rule nsqn_invalidate_other)
+        with \<open>nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip\<close> show ?thesis by simp
+      qed
+      with \<open>dip\<in>kD(rt (\<sigma>' (nhop dip)))\<close>
+        show "dip \<in> kD (rt (\<sigma>' (nhop dip)))
+              \<and> nsqn (invalidate (rt (\<sigma> i)) (dests (\<sigma> i))) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip" ..
+    qed
+  } note basic_prerr = this
+
+  { fix \<sigma> \<sigma>' :: "ip \<Rightarrow> state"
+    assume a1: "\<forall>dip\<in>kD(rt (\<sigma> i)). the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                \<longrightarrow> dip\<in>kD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                    \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))) dip"
+       and a2: "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+    have "\<forall>dip\<in>kD(rt (\<sigma> i)).
+          the (nhop (update (rt (\<sigma> i)) (sip (\<sigma> i)) (0, unk, val, Suc 0, sip (\<sigma> i), {})) dip) \<noteq> dip \<longrightarrow>
+          dip\<in>kD(rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) (sip (\<sigma> i))
+                                        (0, unk, val, Suc 0, sip (\<sigma> i), {}))
+                                  dip)))) \<and>
+          nsqn (update (rt (\<sigma> i)) (sip (\<sigma> i)) (0, unk, val, Suc 0, sip (\<sigma> i), {})) dip
+          \<le> nsqn (rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) (sip (\<sigma> i))
+                                      (0, unk, val, Suc 0, sip (\<sigma> i), {}))
+                                dip))))
+             dip" (is "\<forall>dip\<in>kD(rt (\<sigma> i)). ?P dip")
+    proof
+      fix dip
+      assume "dip\<in>kD(rt (\<sigma> i))"
+      with a1 and a2  
+        have "the (nhop (rt (\<sigma> i)) dip) \<noteq> dip \<longrightarrow> dip\<in>kD(rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip))))
+                        \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip)))) dip"
+           by - (drule(1) basic, auto)
+      thus "?P dip" by (cases "dip = sip (\<sigma> i)") auto
+    qed
+  } note nhop_update_sip = this
+
+  { fix \<sigma> \<sigma>' oip sip osn hops
+    assume pre: "\<forall>dip\<in>kD (rt (\<sigma> i)). the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                 \<longrightarrow> dip\<in>kD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                     \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))) dip"
+       and qinc: "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+       and *: "sip \<noteq> oip \<longrightarrow> oip\<in>kD(rt (\<sigma> sip))
+                             \<and> osn \<le> nsqn (rt (\<sigma> sip)) oip
+                             \<and> (nsqn (rt (\<sigma> sip)) oip = osn
+                                \<longrightarrow> the (dhops (rt (\<sigma> sip)) oip) \<le> hops
+                                    \<or> the (flag (rt (\<sigma> sip)) oip) = inv)"
+    from pre and qinc
+      have pre': "\<forall>dip\<in>kD (rt (\<sigma> i)). the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                   \<longrightarrow> dip\<in>kD(rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip))))
+                       \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip)))) dip"
+        by (rule basic)
+    have "(the (nhop (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})) oip) \<noteq> oip
+           \<longrightarrow> oip\<in>kD(rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) oip
+                                                   (osn, kno, val, Suc hops, sip, {})) oip))))
+                \<and> nsqn (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})) oip
+                   \<le> nsqn (rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) oip
+                                                   (osn, kno, val, Suc hops, sip, {})) oip)))) oip)"
+       (is "?nhop_not_oip \<longrightarrow> ?oip_in_kD \<and> ?nsqn_le_nsqn")
+     proof (rule, split update_rt_split_asm)
+       assume "rt (\<sigma> i) = update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})"
+          and "the (nhop (rt (\<sigma> i)) oip) \<noteq> oip"
+       with pre' show "?oip_in_kD \<and> ?nsqn_le_nsqn" by auto
+     next
+       assume rtnot: "rt (\<sigma> i) \<noteq> update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})"
+          and notoip: ?nhop_not_oip
+       with * qinc have ?oip_in_kD
+         by auto
+       moreover with * pre qinc rtnot notoip have ?nsqn_le_nsqn
+        by simp (metis kD_nsqn_quality_increases_trans)
+       ultimately show "?oip_in_kD \<and> ?nsqn_le_nsqn" ..
+     qed
+  } note update1 = this
+
+  { fix \<sigma> \<sigma>' oip sip osn hops
+    assume pre: "\<forall>dip\<in>kD (rt (\<sigma> i)). the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                  \<longrightarrow> dip\<in>kD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                      \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))) dip"
+       and qinc: "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+       and *: "sip \<noteq> oip \<longrightarrow> oip\<in>kD(rt (\<sigma> sip))
+                           \<and> osn \<le> nsqn (rt (\<sigma> sip)) oip
+                           \<and> (nsqn (rt (\<sigma> sip)) oip = osn
+                              \<longrightarrow> the (dhops (rt (\<sigma> sip)) oip) \<le> hops
+                                  \<or> the (flag (rt (\<sigma> sip)) oip) = inv)"
+    from pre and qinc
+      have pre': "\<forall>dip\<in>kD (rt (\<sigma> i)). the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                   \<longrightarrow> dip\<in>kD(rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip))))
+                       \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip)))) dip"
+        by (rule basic)
+    have "\<forall>dip\<in>kD(rt (\<sigma> i)).
+           the (nhop (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})) dip) \<noteq> dip
+           \<longrightarrow> dip\<in>kD(rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) oip
+                                                  (osn, kno, val, Suc hops, sip, {})) dip))))
+               \<and> nsqn (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})) dip
+                  \<le> nsqn (rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) oip
+                                                  (osn, kno, val, Suc hops, sip, {})) dip)))) dip"
+       (is "\<forall>dip\<in>kD(rt (\<sigma> i)). _ \<longrightarrow> ?dip_in_kD dip \<and> ?nsqn_le_nsqn dip")
+     proof (intro ballI impI, split update_rt_split_asm)
+       fix dip
+       assume "dip\<in>kD(rt (\<sigma> i))"
+          and "the (nhop (rt (\<sigma> i)) dip) \<noteq> dip"
+          and "rt (\<sigma> i) = update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})"
+        with pre' show "?dip_in_kD dip \<and> ?nsqn_le_nsqn dip" by simp
+     next
+       fix dip
+       assume "dip\<in>kD(rt (\<sigma> i))"
+          and notdip: "the (nhop (update (rt (\<sigma> i)) oip
+                             (osn, kno, val, Suc hops, sip, {})) dip) \<noteq> dip"
+          and rtnot: "rt (\<sigma> i) \<noteq> update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})"
+       show "?dip_in_kD dip \<and> ?nsqn_le_nsqn dip"
+       proof (cases "dip = oip")
+         assume "dip \<noteq> oip"
+         with pre' \<open>dip\<in>kD(rt (\<sigma> i))\<close> notdip
+           show ?thesis by clarsimp
+       next
+         assume "dip = oip"
+         with rtnot qinc \<open>dip\<in>kD(rt (\<sigma> i))\<close> notdip *
+           have "?dip_in_kD dip"
+            by simp (metis kD_quality_increases)
+         moreover from \<open>dip = oip\<close> rtnot qinc \<open>dip\<in>kD(rt (\<sigma> i))\<close> notdip *
+           have "?nsqn_le_nsqn dip" by simp (metis kD_nsqn_quality_increases_trans)
+         ultimately show ?thesis ..
+       qed
+     qed
+  } note update2 = this
+
+  have "opaodv i \<Turnstile> (?S, ?U \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<sigma>, _).
+                   \<forall>dip \<in> kD(rt (\<sigma> i)). the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                      \<longrightarrow> dip \<in> kD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                          \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))) dip)"
+    by (inv_cterms inv add: oseq_step_invariant_sterms [OF oquality_increases_nsqn_fresh aodv_wf
+                                                           oaodv_trans]
+                            onl_oinvariant_sterms [OF aodv_wf oaddpreRT_welldefined]
+                            onl_oinvariant_sterms [OF aodv_wf odests_vD_inc_sqn]
+                            onl_oinvariant_sterms [OF aodv_wf oprerr_guard]
+                            onl_oinvariant_sterms [OF aodv_wf rreq_sip]
+                            onl_oinvariant_sterms [OF aodv_wf rrep_sip]
+                            onl_oinvariant_sterms [OF aodv_wf rerr_sip]
+                            other_quality_increases
+                            other_localD
+                      solve: basic basic_prerr
+                      simp add: seqlsimp nsqn_invalidate nhop_update_sip
+                      simp del: One_nat_def)
+       (rule conjI, erule(2) update1, erule(2) update2)+
+
+    thus ?thesis unfolding Let_def by auto
+  qed
+
+text \<open>Proposition 7.30\<close>
+
+lemmas okD_unk_or_atleast_one =
+         open_seq_invariant [OF kD_unk_or_atleast_one initiali_aodv,
+                             simplified seql_onl_swap]
+
+lemmas ozero_seq_unk_hops_one =
+         open_seq_invariant [OF zero_seq_unk_hops_one initiali_aodv,
+                             simplified seql_onl_swap]
+
+lemma oreachable_fresh_okD_unk_or_atleast_one:
+    fixes dip
+  assumes "(\<sigma>, p) \<in> oreachable (opaodv i)
+                       (otherwith ((=)) {i} (orecvmsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m
+                                                             \<and> msg_zhops m)))
+                       (other quality_increases {i})"
+      and "dip\<in>kD(rt (\<sigma> i))"
+    shows "\<pi>\<^sub>3(the (rt (\<sigma> i) dip)) = unk \<or> 1 \<le> \<pi>\<^sub>2(the (rt (\<sigma> i) dip))"
+    (is "?P dip")
+  proof -
+    have "\<exists>l. l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p" by (metis aodv_ex_label)
+    with assms(1) have "\<forall>dip\<in>kD (rt (\<sigma> i)). ?P dip"
+      by - (drule oinvariant_weakenD [OF okD_unk_or_atleast_one [OF oaodv_trans aodv_trans]],
+            auto dest!: otherwith_actionD onlD simp: seqlsimp)
+    with \<open>dip\<in>kD(rt (\<sigma> i))\<close> show ?thesis by simp
+  qed
+
+lemma oreachable_fresh_ozero_seq_unk_hops_one:
+    fixes dip
+  assumes "(\<sigma>, p) \<in> oreachable (opaodv i)
+                     (otherwith ((=)) {i} (orecvmsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m
+                                                         \<and> msg_zhops m)))
+                     (other quality_increases {i})"
+      and "dip\<in>kD(rt (\<sigma> i))"
+    shows "sqn (rt (\<sigma> i)) dip = 0 \<longrightarrow>
+             sqnf (rt (\<sigma> i)) dip = unk
+             \<and> the (dhops (rt (\<sigma> i)) dip) = 1
+             \<and> the (nhop (rt (\<sigma> i)) dip) = dip"
+    (is "?P dip")
+  proof -
+    have "\<exists>l. l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p" by (metis aodv_ex_label)
+    with assms(1) have "\<forall>dip\<in>kD (rt (\<sigma> i)). ?P dip"
+      by - (drule oinvariant_weakenD [OF ozero_seq_unk_hops_one [OF oaodv_trans aodv_trans]],
+            auto dest!: onlD otherwith_actionD simp: seqlsimp)
+    with \<open>dip\<in>kD(rt (\<sigma> i))\<close> show ?thesis by simp
+  qed
+
+lemma seq_nhop_quality_increases':
+  shows "opaodv i \<Turnstile> (otherwith ((=)) {i}
+                        (orecvmsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)),
+                         other quality_increases {i} \<rightarrow>)
+                        onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<sigma>, _). \<forall>dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                               in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip))
+                                                  \<and> nhip \<noteq> dip
+                                                  \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+  (is "_ \<Turnstile> (?S i, _ \<rightarrow>) _")
+  proof -
+    have weaken:
+      "\<And>p I Q R P. p \<Turnstile> (otherwith quality_increases I (orecvmsg Q), other quality_increases I \<rightarrow>) P
+       \<Longrightarrow> p \<Turnstile> (otherwith ((=)) I (orecvmsg (\<lambda>\<sigma> m. Q \<sigma> m \<and> R \<sigma> m)), other quality_increases I \<rightarrow>) P"
+        by auto
+    {
+      fix i a and \<sigma> \<sigma>' :: "ip \<Rightarrow> state"
+      assume a1: "\<forall>dip. dip\<in>vD(rt (\<sigma> i))
+                        \<and> dip\<in>vD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                        \<and> (the (nhop (rt (\<sigma> i)) dip)) \<noteq> dip
+                         \<longrightarrow> rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))"
+         and ow: "?S i \<sigma> \<sigma>' a"
+      have "\<forall>dip. dip\<in>vD(rt (\<sigma> i))
+                  \<and> dip\<in>vD (rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip))))
+                  \<and> (the (nhop (rt (\<sigma> i)) dip)) \<noteq> dip
+               \<longrightarrow> rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip)))"
+      proof clarify
+        fix dip
+        assume a2: "dip\<in>vD(rt (\<sigma> i))"
+           and a3: "dip\<in>vD (rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip))))"
+           and a4: "(the (nhop (rt (\<sigma> i)) dip)) \<noteq> dip"
+        from ow have "\<forall>j. j \<noteq> i \<longrightarrow> \<sigma> j = \<sigma>' j" by auto
+        show "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip)))"
+        proof (cases "(the (nhop (rt (\<sigma> i)) dip)) = i")
+          assume "(the (nhop (rt (\<sigma> i)) dip)) = i"
+          with \<open>dip \<in> vD(rt (\<sigma> i))\<close> have "dip \<in> vD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))" by simp
+          with a1 a2 a4 have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))" by simp
+          with \<open>(the (nhop (rt (\<sigma> i)) dip)) = i\<close> have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> i)" by simp
+          hence False by simp
+          thus ?thesis ..
+        next
+          assume "(the (nhop (rt (\<sigma> i)) dip)) \<noteq> i"
+          with \<open>\<forall>j. j \<noteq> i \<longrightarrow> \<sigma> j = \<sigma>' j\<close>
+            have *: "\<sigma> (the (nhop (rt (\<sigma> i)) dip)) = \<sigma>' (the (nhop (rt (\<sigma> i)) dip))" by simp
+          with \<open>dip\<in>vD (rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip))))\<close>
+            have "dip\<in>vD (rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))" by simp
+          with a1 a2 a4 have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))" by simp
+          with * show ?thesis by simp
+        qed
+      qed
+    } note basic = this
+
+    { fix \<sigma> \<sigma>' a dip sip i
+      assume a1: "\<forall>dip. dip\<in>vD(rt (\<sigma> i))
+                      \<and> dip\<in>vD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                      \<and> the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                      \<longrightarrow> rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))"
+         and ow: "?S i \<sigma> \<sigma>' a"
+      have "\<forall>dip. dip\<in>vD(update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {}))
+           \<and> dip\<in>vD(rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})) dip))))
+           \<and> the (nhop (update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})) dip) \<noteq> dip
+           \<longrightarrow> update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})
+               \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})) dip)))"
+      proof clarify
+        fix dip
+        assume a2: "dip\<in>vD (update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {}))"
+           and a3: "dip\<in>vD(rt (\<sigma>' (the (nhop
+                         (update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})) dip))))"
+           and a4: "the (nhop (update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})) dip) \<noteq> dip"
+        show "update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})
+              \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})) dip)))"
+        proof (cases "dip = sip")
+          assume "dip = sip"
+          with \<open>the (nhop (update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})) dip) \<noteq> dip\<close>
+          have False by simp
+          thus ?thesis ..
+        next
+          assume [simp]: "dip \<noteq> sip"
+          from a2 have "dip\<in>vD(rt (\<sigma> i)) \<or> dip = sip"
+            by (rule vD_update_val)
+          with \<open>dip \<noteq> sip\<close> have "dip\<in>vD(rt (\<sigma> i))" by simp
+          moreover from a3 have "dip\<in>vD(rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip))))" by simp
+          moreover from a4 have "the (nhop (rt (\<sigma> i)) dip) \<noteq> dip" by simp
+          ultimately have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip)))"
+            using a1 ow by - (drule(1) basic, simp)
+          with \<open>dip \<noteq> sip\<close> show ?thesis
+            by - (erule rt_strictly_fresher_update_other, simp)
+        qed
+      qed
+    } note update_0_unk = this
+
+    { fix \<sigma> a \<sigma>' nhop
+      assume pre: "\<forall>dip. dip\<in>vD(rt (\<sigma> i)) \<and> dip\<in>vD(rt (\<sigma> (nhop dip))) \<and> nhop dip \<noteq> dip
+                         \<longrightarrow> rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (nhop dip))"
+         and ow: "?S i \<sigma> \<sigma>' a"
+      have "\<forall>dip. dip \<in> vD (invalidate (rt (\<sigma> i)) (dests (\<sigma> i)))
+                  \<and> dip \<in> vD (rt (\<sigma>' (nhop dip))) \<and> nhop dip \<noteq> dip
+                  \<longrightarrow> rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (nhop dip))"
+      proof clarify
+        fix dip
+        assume "dip\<in>vD(invalidate (rt (\<sigma> i)) (dests (\<sigma> i)))"
+           and "dip\<in>vD(rt (\<sigma>' (nhop dip)))"
+           and "nhop dip \<noteq> dip"
+        from this(1) have "dip\<in>vD (rt (\<sigma> i))"
+          by (clarsimp dest!: vD_invalidate_vD_not_dests)
+        moreover from ow have "\<forall>j. j \<noteq> i \<longrightarrow> \<sigma> j = \<sigma>' j" by auto
+        ultimately have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (nhop dip))"
+          using pre \<open>dip \<in> vD (rt (\<sigma>' (nhop dip)))\<close> \<open>nhop dip \<noteq> dip\<close>
+          by metis
+        with \<open>\<forall>j. j \<noteq> i \<longrightarrow> \<sigma> j = \<sigma>' j\<close> show  "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (nhop dip))"
+          by (metis rt_strictly_fresher_irefl)
+      qed
+    } note invalidate = this
+
+    { fix \<sigma> a \<sigma>' dip oip osn sip hops i
+      assume pre: "\<forall>dip. dip \<in> vD (rt (\<sigma> i))
+                       \<and> dip \<in> vD (rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                       \<and> the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                   \<longrightarrow> rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))"
+         and ow: "?S i \<sigma> \<sigma>' a"
+         and "Suc 0 \<le> osn"
+         and a6: "sip \<noteq> oip \<longrightarrow> oip \<in> kD (rt (\<sigma> sip))
+                                 \<and> osn \<le> nsqn (rt (\<sigma> sip)) oip
+                                 \<and> (nsqn (rt (\<sigma> sip)) oip = osn
+                                    \<longrightarrow> the (dhops (rt (\<sigma> sip)) oip) \<le> hops
+                                         \<or> the (flag (rt (\<sigma> sip)) oip) = inv)"
+         and after: "\<sigma>' i = \<sigma> i\<lparr>rt := update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})\<rparr>"
+      have "\<forall>dip. dip \<in> vD (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {}))
+                \<and> dip \<in> vD (rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) oip
+                                      (osn, kno, val, Suc hops, sip, {})) dip))))
+                \<and> the (nhop (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})) dip) \<noteq> dip
+             \<longrightarrow> update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})
+                 \<sqsubset>\<^bsub>dip\<^esub>
+                 rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})) dip)))"
+      proof clarify
+        fix dip
+        assume a2: "dip\<in>vD(update (rt (\<sigma> i)) oip (osn, kno, val, Suc (hops), sip, {}))"
+           and a3: "dip\<in>vD(rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) oip
+                                                        (osn, kno, val, Suc hops, sip, {})) dip))))"
+           and a4: "the (nhop (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})) dip) \<noteq> dip"
+        from ow have a5: "\<forall>j. j \<noteq> i \<longrightarrow> \<sigma> j = \<sigma>' j" by auto
+        show "update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})
+              \<sqsubset>\<^bsub>dip\<^esub>
+              rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})) dip)))"
+          (is "?rt1 \<sqsubset>\<^bsub>dip\<^esub> ?rt2 dip")
+        proof (cases "?rt1 = rt (\<sigma> i)")
+          assume nochange [simp]:
+            "update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {}) = rt (\<sigma> i)"
+
+          from after have "\<sigma>' i = \<sigma> i" by simp
+          with a5 have "\<forall>j. \<sigma> j = \<sigma>' j" by metis
+
+          from a2 have "dip\<in>vD (rt (\<sigma> i))" by simp
+          moreover from a3 have "dip\<in>vD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))"
+            using nochange and \<open>\<forall>j. \<sigma> j = \<sigma>' j\<close> by clarsimp
+          moreover from a4 have "the (nhop (rt (\<sigma> i)) dip) \<noteq> dip" by simp
+          ultimately have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))"
+            using pre by simp
+
+          hence "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip)))"
+            using \<open>\<forall>j. \<sigma> j = \<sigma>' j\<close> by simp
+          thus "?thesis" by simp
+        next
+          assume change: "?rt1 \<noteq> rt (\<sigma> i)"
+          from after a2 have "dip\<in>kD(rt (\<sigma>' i))" by auto
+          show ?thesis
+          proof (cases "dip = oip")
+            assume "dip \<noteq> oip"
+
+            with a2 have "dip\<in>vD (rt (\<sigma> i))" by auto
+            moreover with a3 a5 after and \<open>dip \<noteq> oip\<close>
+              have "dip\<in>vD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))"
+                by simp metis
+            moreover from a4 and \<open>dip \<noteq> oip\<close> have "the (nhop (rt (\<sigma> i)) dip) \<noteq> dip" by simp
+            ultimately have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))"
+              using pre by simp
+
+            with after and a5 and \<open>dip \<noteq> oip\<close> show ?thesis
+              by simp (metis rt_strictly_fresher_update_other
+                             rt_strictly_fresher_irefl)
+          next
+            assume "dip = oip"
+
+            with a4 and change have "sip \<noteq> oip" by simp
+            with a6 have "oip\<in>kD(rt (\<sigma> sip))"
+                     and "osn \<le> nsqn (rt (\<sigma> sip)) oip" by auto
+
+            from a3 change \<open>dip = oip\<close> have "oip\<in>vD(rt (\<sigma>' sip))" by simp
+            hence "the (flag (rt (\<sigma>' sip)) oip) = val" by simp
+
+            from \<open>oip\<in>kD(rt (\<sigma> sip))\<close>
+            have "osn < nsqn (rt (\<sigma>' sip)) oip \<or> (osn = nsqn (rt (\<sigma>' sip)) oip
+                                                   \<and> the (dhops (rt (\<sigma>' sip)) oip) \<le> hops)"
+            proof
+              assume "oip\<in>vD(rt (\<sigma> sip))"
+              hence "the (flag (rt (\<sigma> sip)) oip) = val" by simp
+              with a6 \<open>sip \<noteq> oip\<close> have "nsqn (rt (\<sigma> sip)) oip = osn \<longrightarrow>
+                                          the (dhops (rt (\<sigma> sip)) oip) \<le> hops"
+                by simp
+              show ?thesis
+              proof (cases "sip = i")
+                assume "sip \<noteq> i"
+                with a5 have "\<sigma> sip = \<sigma>' sip" by simp
+                with \<open>osn \<le> nsqn (rt (\<sigma> sip)) oip\<close>
+                 and \<open>nsqn (rt (\<sigma> sip)) oip = osn \<longrightarrow> the (dhops (rt (\<sigma> sip)) oip) \<le> hops\<close>
+                show ?thesis by auto
+              next
+                \<comment> \<open>alternative to using @{text sip_not_ip}\<close>
+                assume [simp]: "sip = i"
+                have "?rt1 = rt (\<sigma> i)"
+                proof (rule update_cases_kD, simp_all)
+                  from \<open>Suc 0 \<le> osn\<close> show "0 < osn" by simp
+                next
+                  from \<open>oip\<in>kD(rt (\<sigma> sip))\<close> and \<open>sip = i\<close> show "oip\<in>kD(rt (\<sigma> i))"
+                    by simp
+                next
+                  assume "sqn (rt (\<sigma> i)) oip < osn"
+                  also from \<open>osn \<le> nsqn (rt (\<sigma> sip)) oip\<close>
+                    have "... \<le> nsqn (rt (\<sigma> i)) oip" by simp
+                  also have "... \<le> sqn (rt (\<sigma> i)) oip"
+                    by (rule nsqn_sqn)
+                  finally have "sqn (rt (\<sigma> i)) oip < sqn (rt (\<sigma> i)) oip" .
+                  hence False by simp
+                  thus "(\<lambda>a. if a = oip
+                             then Some (osn, kno, val, Suc hops, i, \<pi>\<^sub>7 (the (rt (\<sigma> i) oip)))
+                             else rt (\<sigma> i) a) = rt (\<sigma> i)" ..
+                next
+                  assume "sqn (rt (\<sigma> i)) oip = osn"
+                     and "Suc hops < the (dhops (rt (\<sigma> i)) oip)"
+                  from this(1) and \<open>oip \<in> vD (rt (\<sigma> sip))\<close> have "nsqn (rt (\<sigma> i)) oip = osn"
+                    by simp
+                  with \<open>nsqn (rt (\<sigma> sip)) oip = osn \<longrightarrow> the (dhops (rt (\<sigma> sip)) oip) \<le> hops\<close>
+                    have "the (dhops (rt (\<sigma> i)) oip) \<le> hops" by simp
+                  with \<open>Suc hops < the (dhops (rt (\<sigma> i)) oip)\<close> have False by simp
+                  thus "(\<lambda>a. if a = oip
+                             then Some (osn, kno, val, Suc hops, i, \<pi>\<^sub>7 (the (rt (\<sigma> i) oip)))
+                             else rt (\<sigma> i) a) = rt (\<sigma> i)" ..
+                next
+                  assume "the (flag (rt (\<sigma> i)) oip) = inv"
+                  with \<open>the (flag (rt (\<sigma> sip)) oip) = val\<close> have False by simp
+                  thus "(\<lambda>a. if a = oip
+                             then Some (osn, kno, val, Suc hops, i, \<pi>\<^sub>7 (the (rt (\<sigma> i) oip)))
+                             else rt (\<sigma> i) a) = rt (\<sigma> i)" ..
+                next
+                  from \<open>oip\<in>kD(rt (\<sigma> sip))\<close>
+                    show "(\<lambda>a. if a = oip then Some (the (rt (\<sigma> i) oip)) else rt (\<sigma> i) a) = rt (\<sigma> i)"
+                      by (auto dest!: kD_Some)
+                qed
+                with change have False ..
+                thus ?thesis ..
+              qed
+            next
+              assume "oip\<in>iD(rt (\<sigma> sip))"
+              with \<open>the (flag (rt (\<sigma>' sip)) oip) = val\<close> and a5 have "sip = i"
+                by (metis f.distinct(1) iD_flag_is_inv)
+              from \<open>oip\<in>iD(rt (\<sigma> sip))\<close> have "the (flag (rt (\<sigma> sip)) oip) = inv" by auto
+              with \<open>sip = i\<close> \<open>Suc 0 \<le> osn\<close> change after \<open>oip\<in>kD(rt (\<sigma> sip))\<close>
+                have "nsqn (rt (\<sigma> sip)) oip < nsqn (rt (\<sigma>' sip)) oip"
+                  unfolding update_def
+                  by (clarsimp split: option.split_asm if_split_asm)
+                     (auto simp: sqn_def)
+              with \<open>osn \<le> nsqn (rt (\<sigma> sip)) oip\<close> have "osn < nsqn (rt (\<sigma>' sip)) oip"
+                by simp
+              thus ?thesis ..
+            qed
+            thus ?thesis
+            proof
+              assume osnlt: "osn < nsqn (rt (\<sigma>' sip)) oip"
+              from \<open>dip\<in>kD(rt (\<sigma>' i))\<close> and \<open>dip = oip\<close> have "dip \<in> kD (?rt1)" by simp
+              moreover from a3 have "dip \<in> kD(?rt2 dip)" by simp
+              moreover have "nsqn ?rt1 dip < nsqn (?rt2 dip) dip"
+                proof -
+                  have "nsqn ?rt1 oip = osn"
+                    by (simp add: \<open>dip = oip\<close> nsqn_update_changed_kno_val [OF change [THEN not_sym]])
+                  also have "... < nsqn (rt (\<sigma>' sip)) oip" using osnlt .
+                  also have  "... = nsqn (?rt2 oip) oip" by (simp add: change)
+                  finally show ?thesis
+                    using \<open>dip = oip\<close> by simp
+                qed
+              ultimately show ?thesis
+                by (rule rt_strictly_fresher_ltI)
+            next
+              assume osneq: "osn = nsqn (rt (\<sigma>' sip)) oip \<and> the (dhops (rt (\<sigma>' sip)) oip) \<le> hops"
+
+              have "oip\<in>kD(?rt1)" by simp
+              moreover from a3 \<open>dip = oip\<close> have "oip\<in>kD(?rt2 oip)" by simp
+
+              moreover have "nsqn ?rt1 oip = nsqn (?rt2 oip) oip"
+              proof -
+                from osneq have "osn = nsqn (rt (\<sigma>' sip)) oip" ..
+                also have "osn = nsqn ?rt1 oip"
+                  by (simp add: \<open>dip = oip\<close> nsqn_update_changed_kno_val [OF change [THEN not_sym]])
+                also have  "nsqn (rt (\<sigma>' sip)) oip = nsqn (?rt2 oip) oip"
+                  by (simp add: change)
+                finally show ?thesis .
+              qed
+
+              moreover have "\<pi>\<^sub>5(the (?rt2 oip oip)) < \<pi>\<^sub>5(the (?rt1 oip))"
+              proof -
+                from osneq have "the (dhops (rt (\<sigma>' sip)) oip) \<le> hops" ..
+                moreover from \<open>oip \<in> vD (rt (\<sigma>' sip))\<close> have "oip\<in>kD(rt (\<sigma>' sip))" by auto
+                ultimately have "\<pi>\<^sub>5(the (rt (\<sigma>' sip) oip)) \<le> hops"
+                  by (auto simp add: proj5_eq_dhops)
+                also from change after have "hops < \<pi>\<^sub>5(the (rt (\<sigma>' i) oip))"
+                  by (simp add: proj5_eq_dhops) (metis dhops_update_changed lessI)
+                finally have "\<pi>\<^sub>5(the (rt (\<sigma>' sip) oip)) < \<pi>\<^sub>5(the (rt (\<sigma>' i) oip))" .
+                with change after show ?thesis by simp
+              qed
+
+              ultimately have "?rt1 \<sqsubset>\<^bsub>oip\<^esub> ?rt2 oip"
+                by (rule rt_strictly_fresher_eqI)
+              with \<open>dip = oip\<close> show ?thesis by simp
+                qed
+          qed
+       qed
+     qed
+    } note rreq_rrep_update = this
+
+    have "opaodv i \<Turnstile> (otherwith ((=)) {i} (orecvmsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m
+                                                            \<and> msg_zhops m)),
+                       other quality_increases {i} \<rightarrow>)
+            onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V
+           (\<lambda>(\<sigma>, _). \<forall>dip. dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                           \<and> the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                           \<longrightarrow> rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))"
+      proof (inv_cterms inv add: onl_oinvariant_sterms [OF aodv_wf rreq_sip [THEN weaken]]
+                                 onl_oinvariant_sterms [OF aodv_wf rrep_sip [THEN weaken]]
+                                 onl_oinvariant_sterms [OF aodv_wf rerr_sip [THEN weaken]]
+                                 onl_oinvariant_sterms [OF aodv_wf oosn_rreq [THEN weaken]]
+                                 onl_oinvariant_sterms [OF aodv_wf odsn_rrep [THEN weaken]]
+                                 onl_oinvariant_sterms [OF aodv_wf oaddpreRT_welldefined]
+                        solve: basic update_0_unk invalidate rreq_rrep_update
+                        simp add: seqlsimp)
+        fix \<sigma> \<sigma>' p l
+        assume or: "(\<sigma>, p) \<in> oreachable (opaodv i) (?S i)  (other quality_increases {i})"
+           and "other quality_increases {i} \<sigma> \<sigma>'"
+           and ll: "l \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p"
+           and pre: "\<forall>dip. dip\<in>vD (rt (\<sigma> i))
+                           \<and> dip\<in>vD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                           \<and> the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                        \<longrightarrow> rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))"
+        from this(1-2)
+          have or': "(\<sigma>', p) \<in> oreachable (opaodv i) (?S i)  (other quality_increases {i})"
+            by - (rule oreachable_other')
+
+        from or and ll have next_hop: "\<forall>dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                             in dip \<in> kD(rt (\<sigma> i)) \<and> nhip \<noteq> dip
+                                             \<longrightarrow> dip \<in> kD(rt (\<sigma> nhip))
+                                                 \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> nhip)) dip"
+          by (auto dest!: onl_oinvariant_weakenD [OF seq_compare_next_hop'])
+
+        from or and ll have unk_hops_one: "\<forall>dip\<in>kD (rt (\<sigma> i)). sqn (rt (\<sigma> i)) dip = 0
+                                                \<longrightarrow> sqnf (rt (\<sigma> i)) dip = unk
+                                                    \<and> the (dhops (rt (\<sigma> i)) dip) = 1
+                                                    \<and> the (nhop (rt (\<sigma> i)) dip) = dip"
+          by (auto dest!: onl_oinvariant_weakenD [OF ozero_seq_unk_hops_one
+                                                     [OF oaodv_trans aodv_trans]]
+                          otherwith_actionD
+                          simp: seqlsimp)
+
+        from \<open>other quality_increases {i} \<sigma> \<sigma>'\<close> have "\<sigma>' i = \<sigma> i" by auto
+        hence "quality_increases (\<sigma> i) (\<sigma>' i)" by auto
+        with \<open>other quality_increases {i} \<sigma> \<sigma>'\<close> have "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+          by - (erule otherE, metis singleton_iff)
+
+        show "\<forall>dip. dip \<in> vD (rt (\<sigma>' i))
+                  \<and> dip \<in> vD (rt (\<sigma>' (the (nhop (rt (\<sigma>' i)) dip))))
+                  \<and> the (nhop (rt (\<sigma>' i)) dip) \<noteq> dip
+              \<longrightarrow> rt (\<sigma>' i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (the (nhop (rt (\<sigma>' i)) dip)))"
+        proof clarify
+          fix dip
+          assume "dip\<in>vD(rt (\<sigma>' i))"
+             and "dip\<in>vD(rt (\<sigma>' (the (nhop (rt (\<sigma>' i)) dip))))"
+             and "the (nhop (rt (\<sigma>' i)) dip) \<noteq> dip"
+          from this(1) and \<open>\<sigma>' i = \<sigma> i\<close> have "dip\<in>vD(rt (\<sigma> i))"
+                                         and "dip\<in>kD(rt (\<sigma> i))"
+            by auto
+
+          from \<open>the (nhop (rt (\<sigma>' i)) dip) \<noteq> dip\<close> and \<open>\<sigma>' i = \<sigma> i\<close>
+            have "the (nhop (rt (\<sigma> i)) dip) \<noteq> dip" (is "?nhip \<noteq> _") by simp
+          with \<open>dip\<in>kD(rt (\<sigma> i))\<close> and next_hop
+            have "dip\<in>kD(rt (\<sigma> (?nhip)))"
+             and nsqns: "nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> ?nhip)) dip"
+               by (auto simp: Let_def)
+
+          have "0 < sqn (rt (\<sigma> i)) dip"
+            proof (rule neq0_conv [THEN iffD1, OF notI])
+              assume "sqn (rt (\<sigma> i)) dip = 0"
+              with \<open>dip\<in>kD(rt (\<sigma> i))\<close> and unk_hops_one
+                have "?nhip = dip" by simp
+              with \<open>?nhip \<noteq> dip\<close> show False ..
+            qed
+          also have "... = nsqn (rt (\<sigma> i)) dip"
+            by (rule vD_nsqn_sqn [OF \<open>dip\<in>vD(rt (\<sigma> i))\<close>, THEN sym])
+          also have "... \<le> nsqn (rt (\<sigma> ?nhip)) dip"
+            by (rule nsqns)
+          also have "... \<le> sqn (rt (\<sigma> ?nhip)) dip"
+            by (rule nsqn_sqn)
+          finally have "0 < sqn (rt (\<sigma> ?nhip)) dip" .
+
+          have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' ?nhip)"
+          proof (cases "dip\<in>vD(rt (\<sigma> ?nhip))")
+            assume "dip\<in>vD(rt (\<sigma> ?nhip))"
+            with pre \<open>dip\<in>vD(rt (\<sigma> i))\<close> and \<open>?nhip \<noteq> dip\<close>
+              have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> ?nhip)" by auto
+            moreover from \<open>\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)\<close>
+              have "quality_increases (\<sigma> ?nhip) (\<sigma>' ?nhip)" ..
+            ultimately show ?thesis
+              using \<open>dip\<in>kD(rt (\<sigma> ?nhip))\<close>
+              by (rule strictly_fresher_quality_increases_right)
+          next
+            assume "dip\<notin>vD(rt (\<sigma> ?nhip))"
+            with \<open>dip\<in>kD(rt (\<sigma> ?nhip))\<close> have "dip\<in>iD(rt (\<sigma> ?nhip))" ..
+            hence "the (flag (rt (\<sigma> ?nhip)) dip) = inv"
+              by auto
+            have "nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> ?nhip)) dip"
+              by (rule nsqns)
+            also from \<open>dip\<in>iD(rt (\<sigma> ?nhip))\<close>
+              have "... = sqn (rt (\<sigma> ?nhip)) dip - 1" ..
+            also have "... < sqn (rt (\<sigma>' ?nhip)) dip"
+              proof -
+                from \<open>\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)\<close>
+                  have "quality_increases (\<sigma> ?nhip) (\<sigma>' ?nhip)" ..
+                hence "\<forall>ip. sqn (rt (\<sigma> ?nhip)) ip \<le> sqn (rt (\<sigma>' ?nhip)) ip" by auto
+                hence "sqn (rt (\<sigma> ?nhip)) dip \<le> sqn (rt (\<sigma>' ?nhip)) dip" ..
+                with \<open>0 < sqn (rt (\<sigma> ?nhip)) dip\<close> show ?thesis by auto
+              qed
+            also have "... = nsqn (rt (\<sigma>' ?nhip)) dip"
+              proof (rule vD_nsqn_sqn [THEN sym])
+                from \<open>dip\<in>vD(rt (\<sigma>' (the (nhop (rt (\<sigma>' i)) dip))))\<close> and \<open>\<sigma>' i = \<sigma> i\<close>
+                  show "dip\<in>vD(rt (\<sigma>' ?nhip))" by simp
+              qed
+            finally have "nsqn (rt (\<sigma> i)) dip < nsqn (rt (\<sigma>' ?nhip)) dip" .
+
+            moreover from \<open>dip\<in>vD(rt (\<sigma>' (the (nhop (rt (\<sigma>' i)) dip))))\<close> and \<open>\<sigma>' i = \<sigma> i\<close>
+              have "dip\<in>kD(rt (\<sigma>' ?nhip))" by auto
+            ultimately show "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' ?nhip)"
+              using \<open>dip\<in>kD(rt (\<sigma> i))\<close> by - (rule rt_strictly_fresher_ltI)
+          qed
+          with \<open>\<sigma>' i = \<sigma> i\<close> show "rt (\<sigma>' i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (the (nhop (rt (\<sigma>' i)) dip)))"
+            by simp
+        qed
+      qed
+    thus ?thesis unfolding Let_def .
+  qed
+
+lemma seq_nhop_quality_increases:
+  shows "opaodv i \<Turnstile> (otherwith ((=)) {i}
+                        (orecvmsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)),
+                         other quality_increases {i} \<rightarrow>)
+                        global (\<lambda>\<sigma>. \<forall>dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                          in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                                             \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+  by (rule oinvariant_weakenE [OF seq_nhop_quality_increases']) (auto dest!: onlD)
+
+end

formal/afp/AODV/Loop_Freedom.thy

@@ -0,0 +1,123 @@
+(*  Title:       Loop_Freedom.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "Routing graphs and loop freedom"
+
+theory Loop_Freedom
+imports Aodv_Predicates Fresher
+begin
+
+text \<open>Define the central theorem that relates an invariant over network states to the absence
+      of loops in the associate routing graph.\<close>
+
+definition
+  rt_graph :: "(ip \<Rightarrow> state) \<Rightarrow> ip \<Rightarrow> ip rel"
+where
+  "rt_graph \<sigma> = (\<lambda>dip.
+     {(ip, ip') | ip ip' dsn dsk hops pre.
+        ip \<noteq> dip \<and> rt (\<sigma> ip) dip = Some (dsn, dsk, val, hops, ip', pre)})"
+
+text \<open>Given the state of a network @{term \<sigma>}, a routing graph for a given destination
+      ip address @{term dip} abstracts the details of routing tables into nodes
+      (ip addresses) and vertices (valid routes between ip addresses).\<close>
+
+lemma rt_graphE [elim]:
+    fixes n dip ip ip'
+  assumes "(ip, ip') \<in> rt_graph \<sigma> dip"
+    shows "ip \<noteq> dip \<and> (\<exists>r. rt (\<sigma> ip) = r
+                            \<and> (\<exists>dsn dsk hops pre. r dip = Some (dsn, dsk, val, hops, ip', pre)))"
+  using assms unfolding rt_graph_def by auto
+
+lemma rt_graph_vD [dest]:
+  "\<And>ip ip' \<sigma> dip. (ip, ip') \<in> rt_graph \<sigma> dip \<Longrightarrow> dip \<in> vD(rt (\<sigma> ip))"
+  unfolding rt_graph_def vD_def by auto
+
+lemma rt_graph_vD_trans [dest]:
+  "\<And>ip ip' \<sigma> dip. (ip, ip') \<in> (rt_graph \<sigma> dip)\<^sup>+ \<Longrightarrow> dip \<in> vD(rt (\<sigma> ip))"
+  by (erule converse_tranclE) auto
+
+lemma rt_graph_not_dip [dest]:
+  "\<And>ip ip' \<sigma> dip. (ip, ip') \<in> rt_graph \<sigma> dip \<Longrightarrow> ip \<noteq> dip"
+  unfolding rt_graph_def by auto
+
+lemma rt_graph_not_dip_trans [dest]:
+  "\<And>ip ip' \<sigma> dip. (ip, ip') \<in> (rt_graph \<sigma> dip)\<^sup>+ \<Longrightarrow> ip \<noteq> dip"
+  by (erule converse_tranclE) auto
+
+text "NB: the property below cannot be lifted to the transitive closure"
+
+lemma rt_graph_nhip_is_nhop [dest]:
+  "\<And>ip ip' \<sigma> dip. (ip, ip') \<in> rt_graph \<sigma> dip \<Longrightarrow> ip' = the (nhop (rt (\<sigma> ip)) dip)"
+  unfolding rt_graph_def by auto
+
+theorem inv_to_loop_freedom:
+  assumes "\<forall>i dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                   in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                      \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip))"
+    shows "\<forall>dip. irrefl ((rt_graph \<sigma> dip)\<^sup>+)"
+  using assms proof (intro allI)
+    fix \<sigma> :: "ip \<Rightarrow> state" and dip
+    assume inv: "\<forall>ip dip.
+                  let nhip = the (nhop (rt (\<sigma> ip)) dip)
+                  in dip \<in> vD(rt (\<sigma> ip)) \<inter> vD(rt (\<sigma> nhip)) \<and>
+                     nhip \<noteq> dip \<longrightarrow> rt (\<sigma> ip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip)"
+    { fix ip ip'
+      assume "(ip, ip') \<in> (rt_graph \<sigma> dip)\<^sup>+"
+         and "dip \<in> vD(rt (\<sigma> ip'))"
+         and "ip' \<noteq> dip"
+       hence "rt (\<sigma> ip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> ip')"
+         proof induction
+           fix nhip
+           assume "(ip, nhip) \<in> rt_graph \<sigma> dip"
+              and "dip \<in> vD(rt (\<sigma> nhip))"
+              and "nhip \<noteq> dip"
+           from \<open>(ip, nhip) \<in> rt_graph \<sigma> dip\<close> have "dip \<in> vD(rt (\<sigma> ip))"
+                                               and "nhip = the (nhop (rt (\<sigma> ip)) dip)"
+             by auto
+           from \<open>dip \<in> vD(rt (\<sigma> ip))\<close> and \<open>dip \<in> vD(rt (\<sigma> nhip))\<close>
+             have "dip \<in> vD(rt (\<sigma> ip)) \<inter> vD(rt (\<sigma> nhip))" ..
+           with \<open>nhip = the (nhop (rt (\<sigma> ip)) dip)\<close>
+                and \<open>nhip \<noteq> dip\<close>
+                and inv
+             show "rt (\<sigma> ip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip)"
+             by (clarsimp simp: Let_def)
+         next
+           fix nhip nhip'
+           assume "(ip, nhip) \<in> (rt_graph \<sigma> dip)\<^sup>+"
+              and "(nhip, nhip') \<in> rt_graph \<sigma> dip"
+              and IH: "\<lbrakk> dip \<in> vD(rt (\<sigma> nhip)); nhip \<noteq> dip \<rbrakk> \<Longrightarrow> rt (\<sigma> ip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip)"
+              and "dip \<in> vD(rt (\<sigma> nhip'))"
+              and "nhip' \<noteq> dip"
+           from \<open>(nhip, nhip') \<in> rt_graph \<sigma> dip\<close> have 1: "dip \<in> vD(rt (\<sigma> nhip))"
+                                                  and 2: "nhip \<noteq> dip"
+                                                  and "nhip' = the (nhop (rt (\<sigma> nhip)) dip)"
+             by auto
+           from 1 2 have "rt (\<sigma> ip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip)" by (rule IH)
+           also have "rt (\<sigma> nhip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip')"
+             proof -
+               from \<open>dip \<in> vD(rt (\<sigma> nhip))\<close> and \<open>dip \<in> vD(rt (\<sigma> nhip'))\<close>
+                 have "dip \<in> vD(rt (\<sigma> nhip)) \<inter> vD(rt (\<sigma> nhip'))" ..
+               with \<open>nhip' \<noteq> dip\<close>
+                    and \<open>nhip' = the (nhop (rt (\<sigma> nhip)) dip)\<close>
+                    and inv
+                 show "rt (\<sigma> nhip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip')"
+                   by (clarsimp simp: Let_def)
+             qed
+           finally show "rt (\<sigma> ip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip')" .
+         qed } note fresher = this
+
+    show "irrefl ((rt_graph \<sigma> dip)\<^sup>+)"
+    unfolding irrefl_def proof (intro allI notI)
+      fix ip
+      assume "(ip, ip) \<in> (rt_graph \<sigma> dip)\<^sup>+"
+      moreover then have "dip \<in> vD(rt (\<sigma> ip))"
+                     and "ip \<noteq> dip"
+        by auto
+      ultimately have "rt (\<sigma> ip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> ip)" by (rule fresher)
+      thus False by simp
+    qed
+  qed
+
+end

formal/afp/AODV/OAodv.thy

@@ -0,0 +1,47 @@
+(*  Title:       OAodv.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "The `open' AODV model"
+
+theory OAodv
+imports Aodv AWN.OAWN_SOS_Labels AWN.OAWN_Convert
+begin
+
+text \<open>Definitions for stating and proving global network properties over individual processes.\<close>
+
+definition \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V' :: "((ip \<Rightarrow> state) \<times> ((state, msg, pseqp, pseqp label) seqp)) set"
+where "\<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V' \<equiv> {(\<lambda>i. aodv_init i, \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PAodv)}"
+
+abbreviation opaodv
+  :: "ip \<Rightarrow> ((ip \<Rightarrow> state) \<times> (state, msg, pseqp, pseqp label) seqp, msg seq_action) automaton"
+where
+  "opaodv i \<equiv> \<lparr> init = \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V', trans = oseqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i \<rparr>"
+
+lemma initiali_aodv [intro!, simp]: "initiali i (init (opaodv i)) (init (paodv i))"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'_def by rule simp_all
+
+lemma oaodv_control_within [simp]: "control_within \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (init (opaodv i))"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'_def by (rule control_withinI) (auto simp del: \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_simps)
+
+lemma \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'_labels [simp]: "(\<sigma>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V' \<Longrightarrow>  labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p = {PAodv-:0}"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'_def by simp
+
+lemma oaodv_init_kD_empty [simp]:
+  "(\<sigma>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V' \<Longrightarrow> kD (rt (\<sigma> i)) = {}"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'_def kD_def by simp
+
+lemma oaodv_init_vD_empty [simp]:
+  "(\<sigma>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V' \<Longrightarrow> vD (rt (\<sigma> i)) = {}"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'_def vD_def by simp
+
+lemma oaodv_trans: "trans (opaodv i) = oseqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i"
+  by simp
+
+declare
+  oseq_invariant_ctermsI [OF aodv_wf oaodv_control_within aodv_simple_labels oaodv_trans, cterms_intros]
+  oseq_step_invariant_ctermsI [OF aodv_wf oaodv_control_within aodv_simple_labels oaodv_trans, cterms_intros]
+
+end
+

formal/afp/AODV/Quality_Increases.thy

@@ -0,0 +1,456 @@
+(*  Title:       Quality_Increases.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "The quality increases predicate"
+
+theory Quality_Increases
+imports Aodv_Predicates Fresher
+begin
+
+definition quality_increases :: "state \<Rightarrow> state \<Rightarrow> bool"
+where "quality_increases \<xi> \<xi>' \<equiv> (\<forall>dip\<in>kD(rt \<xi>). dip \<in> kD(rt \<xi>') \<and> rt \<xi> \<sqsubseteq>\<^bsub>dip\<^esub> rt \<xi>')
+                                               \<and> (\<forall>dip. sqn (rt \<xi>) dip \<le> sqn (rt \<xi>') dip)"
+
+lemma quality_increasesI [intro!]:
+  assumes "\<And>dip. dip \<in> kD(rt \<xi>) \<Longrightarrow> dip \<in> kD(rt \<xi>')"
+      and "\<And>dip. \<lbrakk> dip \<in> kD(rt \<xi>); dip \<in> kD(rt \<xi>') \<rbrakk> \<Longrightarrow> rt \<xi> \<sqsubseteq>\<^bsub>dip\<^esub> rt \<xi>'"          
+      and "\<And>dip. sqn (rt \<xi>) dip \<le> sqn (rt \<xi>') dip"
+    shows "quality_increases \<xi> \<xi>'"
+  unfolding quality_increases_def using assms by clarsimp
+
+lemma quality_increasesE [elim]:
+    fixes dip
+  assumes "quality_increases \<xi> \<xi>'"
+      and "dip\<in>kD(rt \<xi>)"
+      and "\<lbrakk> dip \<in> kD(rt \<xi>'); rt \<xi> \<sqsubseteq>\<^bsub>dip\<^esub> rt \<xi>'; sqn (rt \<xi>) dip \<le> sqn (rt \<xi>') dip \<rbrakk> \<Longrightarrow> R dip \<xi> \<xi>'"
+    shows "R dip \<xi> \<xi>'"
+  using assms unfolding quality_increases_def by clarsimp
+
+lemma quality_increases_rt_fresherD [dest]:
+    fixes ip
+  assumes "quality_increases \<xi> \<xi>'"
+      and "ip\<in>kD(rt \<xi>)"
+    shows "rt \<xi> \<sqsubseteq>\<^bsub>ip\<^esub> rt \<xi>'"
+  using assms by auto
+
+lemma quality_increases_sqnE [elim]:
+    fixes dip
+  assumes "quality_increases \<xi> \<xi>'"
+      and "sqn (rt \<xi>) dip \<le> sqn (rt \<xi>') dip \<Longrightarrow> R dip \<xi> \<xi>'"
+    shows "R dip \<xi> \<xi>'"
+  using assms unfolding quality_increases_def by clarsimp
+
+lemma quality_increases_refl [intro, simp]: "quality_increases \<xi> \<xi>"
+  by rule simp_all
+
+lemma strictly_fresher_quality_increases_right [elim]:
+    fixes \<sigma> \<sigma>' dip
+  assumes "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip)"                       
+      and qinc: "quality_increases (\<sigma> nhip) (\<sigma>' nhip)"
+      and "dip\<in>kD(rt (\<sigma> nhip))"
+    shows "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' nhip)"
+  proof -
+    from qinc have "rt (\<sigma> nhip) \<sqsubseteq>\<^bsub>dip\<^esub> rt (\<sigma>' nhip)" using \<open>dip\<in>kD(rt (\<sigma> nhip))\<close>
+      by auto
+    with \<open>rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip)\<close> show ?thesis ..
+  qed
+
+lemma kD_quality_increases [elim]:
+  assumes "i\<in>kD(rt \<xi>)"
+      and "quality_increases \<xi> \<xi>'"
+    shows "i\<in>kD(rt \<xi>')"
+  using assms by auto
+
+lemma kD_nsqn_quality_increases [elim]:
+  assumes "i\<in>kD(rt \<xi>)"
+      and "quality_increases \<xi> \<xi>'"
+    shows "i\<in>kD(rt \<xi>') \<and> nsqn (rt \<xi>) i \<le> nsqn (rt \<xi>') i"
+  proof -
+    from assms have "i\<in>kD(rt \<xi>')" ..
+    moreover with assms have "rt \<xi> \<sqsubseteq>\<^bsub>i\<^esub> rt \<xi>'" by auto
+    ultimately have "nsqn (rt \<xi>) i \<le> nsqn (rt \<xi>') i"
+      using \<open>i\<in>kD(rt \<xi>)\<close> by - (erule(2) rt_fresher_imp_nsqn_le)
+    with \<open>i\<in>kD(rt \<xi>')\<close> show ?thesis ..
+  qed
+
+lemma nsqn_quality_increases [elim]:
+  assumes "i\<in>kD(rt \<xi>)"
+      and "quality_increases \<xi> \<xi>'"
+    shows "nsqn (rt \<xi>) i \<le> nsqn (rt \<xi>') i"
+  using assms by (rule kD_nsqn_quality_increases [THEN conjunct2])
+
+lemma kD_nsqn_quality_increases_trans [elim]:
+  assumes "i\<in>kD(rt \<xi>)"
+      and "s \<le> nsqn (rt \<xi>) i"
+      and "quality_increases \<xi> \<xi>'"
+    shows "i\<in>kD(rt \<xi>') \<and> s \<le> nsqn (rt \<xi>') i"
+  proof
+    from \<open>i\<in>kD(rt \<xi>)\<close> and \<open>quality_increases \<xi> \<xi>'\<close> show "i\<in>kD(rt \<xi>')" ..
+  next
+    from \<open>i\<in>kD(rt \<xi>)\<close> and \<open>quality_increases \<xi> \<xi>'\<close> have "nsqn (rt \<xi>) i \<le> nsqn (rt \<xi>') i" ..
+    with \<open>s \<le> nsqn (rt \<xi>) i\<close> show "s \<le> nsqn (rt \<xi>') i" by (rule le_trans)
+  qed
+
+lemma nsqn_quality_increases_nsqn_lt_lt [elim]:
+  assumes "i\<in>kD(rt \<xi>)"
+      and "quality_increases \<xi> \<xi>'"
+     and "s < nsqn (rt \<xi>) i"
+    shows "s < nsqn (rt \<xi>') i"
+  proof -
+    from assms(1-2) have "nsqn (rt \<xi>) i \<le> nsqn (rt \<xi>') i" ..
+    with \<open>s < nsqn (rt \<xi>) i\<close> show "s < nsqn (rt \<xi>') i" by simp
+  qed
+
+lemma nsqn_quality_increases_dhops [elim]:
+  assumes "i\<in>kD(rt \<xi>)"
+      and "quality_increases \<xi> \<xi>'"
+      and "nsqn (rt \<xi>) i = nsqn (rt \<xi>') i"
+    shows "the (dhops (rt \<xi>) i) \<ge> the (dhops (rt \<xi>') i)"
+  using assms unfolding quality_increases_def
+  by (clarsimp) (drule(1) bspec, clarsimp simp: rt_fresher_def2)
+
+lemma nsqn_quality_increases_nsqn_eq_le [elim]:
+  assumes "i\<in>kD(rt \<xi>)"
+      and "quality_increases \<xi> \<xi>'"
+      and "s = nsqn (rt \<xi>) i"
+    shows "s < nsqn (rt \<xi>') i \<or> (s = nsqn (rt \<xi>') i \<and> the (dhops (rt \<xi>) i) \<ge> the (dhops (rt \<xi>') i))"
+  using assms by (metis nat_less_le nsqn_quality_increases nsqn_quality_increases_dhops)
+
+lemma quality_increases_rreq_rrep_props [elim]:
+    fixes sn ip hops sip
+  assumes qinc: "quality_increases (\<sigma> sip) (\<sigma>' sip)"
+      and "1 \<le> sn"
+      and *: "ip\<in>kD(rt (\<sigma> sip)) \<and> sn \<le> nsqn (rt (\<sigma> sip)) ip
+                                \<and> (nsqn (rt (\<sigma> sip)) ip = sn
+                                    \<longrightarrow> (the (dhops (rt (\<sigma> sip)) ip) \<le> hops
+                                          \<or> the (flag (rt (\<sigma> sip)) ip) = inv))"
+    shows "ip\<in>kD(rt (\<sigma>' sip)) \<and> sn \<le> nsqn (rt (\<sigma>' sip)) ip
+                              \<and> (nsqn (rt (\<sigma>' sip)) ip = sn
+                                  \<longrightarrow> (the (dhops (rt (\<sigma>' sip)) ip) \<le> hops
+                                        \<or> the (flag (rt (\<sigma>' sip)) ip) = inv))"
+      (is "_ \<and> ?nsqnafter")
+  proof -
+    from *  obtain "ip\<in>kD(rt (\<sigma> sip))" and "sn \<le> nsqn (rt (\<sigma> sip)) ip" by auto
+
+    from \<open>quality_increases (\<sigma> sip) (\<sigma>' sip)\<close>
+       have "sqn (rt (\<sigma> sip)) ip \<le> sqn (rt (\<sigma>' sip)) ip" ..
+    from \<open>quality_increases (\<sigma> sip) (\<sigma>' sip)\<close> and \<open>ip\<in>kD (rt (\<sigma> sip))\<close>
+      have "ip\<in>kD (rt (\<sigma>' sip))" ..
+
+    from \<open>sn \<le> nsqn (rt (\<sigma> sip)) ip\<close> have ?nsqnafter
+    proof
+      assume "sn < nsqn (rt (\<sigma> sip)) ip"
+      also from \<open>ip\<in>kD(rt (\<sigma> sip))\<close> and \<open>quality_increases (\<sigma> sip) (\<sigma>' sip)\<close>
+        have "... \<le> nsqn (rt (\<sigma>' sip)) ip" ..
+      finally have "sn < nsqn (rt (\<sigma>' sip)) ip" .
+      thus ?thesis by simp
+    next
+      assume "sn = nsqn (rt (\<sigma> sip)) ip"
+      with \<open>ip\<in>kD(rt (\<sigma> sip))\<close> and \<open>quality_increases (\<sigma> sip) (\<sigma>' sip)\<close>
+        have "sn < nsqn (rt (\<sigma>' sip)) ip
+              \<or> (sn = nsqn (rt (\<sigma>' sip)) ip
+                 \<and> the (dhops (rt (\<sigma>' sip)) ip) \<le> the (dhops (rt (\<sigma> sip)) ip))" ..
+      hence "sn < nsqn (rt (\<sigma>' sip)) ip
+              \<or> (nsqn (rt (\<sigma>' sip)) ip = sn \<and> (the (dhops (rt (\<sigma>' sip)) ip) \<le> hops
+                                                 \<or> the (flag (rt (\<sigma>' sip)) ip) = inv))"
+      proof
+        assume "sn < nsqn (rt (\<sigma>' sip)) ip" thus ?thesis ..
+      next
+        assume "sn = nsqn (rt (\<sigma>' sip)) ip
+                \<and> the (dhops (rt (\<sigma> sip)) ip) \<ge> the (dhops (rt (\<sigma>' sip)) ip)"
+        hence "sn = nsqn (rt (\<sigma>' sip)) ip"
+          and "the (dhops (rt (\<sigma>' sip)) ip) \<le> the (dhops (rt (\<sigma> sip)) ip)" by auto
+
+        from * and \<open>sn = nsqn (rt (\<sigma> sip)) ip\<close> have "the (dhops (rt (\<sigma> sip)) ip) \<le> hops
+                                                       \<or> the (flag (rt (\<sigma> sip)) ip) = inv"
+          by simp
+        thus ?thesis
+        proof
+          assume "the (dhops (rt (\<sigma> sip)) ip) \<le> hops"
+          with  \<open>the (dhops (rt (\<sigma>' sip)) ip) \<le> the (dhops (rt (\<sigma> sip)) ip)\<close>
+           have "the (dhops (rt (\<sigma>' sip)) ip) \<le> hops" by simp
+          with \<open>sn = nsqn (rt (\<sigma>' sip)) ip\<close> show ?thesis by simp
+        next
+          assume "the (flag (rt (\<sigma> sip)) ip) = inv"
+          with \<open>ip\<in>kD(rt (\<sigma> sip))\<close> have "nsqn (rt (\<sigma> sip)) ip = sqn (rt (\<sigma> sip)) ip - 1" ..
+
+          with \<open>sn \<ge> 1\<close> and \<open>sn = nsqn (rt (\<sigma> sip)) ip\<close>
+            have "sqn (rt (\<sigma> sip)) ip > 1" by simp
+
+          from \<open>ip\<in>kD(rt (\<sigma>' sip))\<close> show ?thesis
+          proof (rule vD_or_iD)
+            assume "ip\<in>iD(rt (\<sigma>' sip))"
+            hence "the (flag (rt (\<sigma>' sip)) ip) = inv" ..
+            with \<open>sn = nsqn (rt (\<sigma>' sip)) ip\<close> show ?thesis
+              by simp
+          next
+            (* the tricky case: sn = nsqn (rt (\<sigma>' sip)) ip
+                                \<and> ip\<in>iD(rt (\<sigma> sip))
+                                \<and> ip\<in>vD(rt (\<sigma>' sip)) *)
+            assume "ip\<in>vD(rt (\<sigma>' sip))"
+            hence "nsqn (rt (\<sigma>' sip)) ip = sqn (rt (\<sigma>' sip)) ip" ..
+            with \<open>sqn (rt (\<sigma> sip)) ip \<le> sqn (rt (\<sigma>' sip)) ip\<close>
+              have "nsqn (rt (\<sigma>' sip)) ip \<ge> sqn (rt (\<sigma> sip)) ip" by simp
+
+            with \<open>sqn (rt (\<sigma> sip)) ip > 1\<close>
+              have "nsqn (rt (\<sigma>' sip)) ip > sqn (rt (\<sigma> sip)) ip - 1" by simp
+            with \<open>nsqn (rt (\<sigma> sip)) ip = sqn (rt (\<sigma> sip)) ip - 1\<close>
+              have "nsqn (rt (\<sigma>' sip)) ip > nsqn (rt (\<sigma> sip)) ip" by simp
+            with \<open>sn = nsqn (rt (\<sigma> sip)) ip\<close> have "nsqn (rt (\<sigma>' sip)) ip > sn"
+              by simp
+            thus ?thesis ..
+          qed
+        qed
+      qed
+      thus ?thesis by (metis (mono_tags) le_cases not_le)
+    qed
+    with \<open>ip\<in>kD (rt (\<sigma>' sip))\<close> show "ip\<in>kD (rt (\<sigma>' sip)) \<and> ?nsqnafter" ..
+  qed
+
+lemma quality_increases_rreq_rrep_props':
+    fixes sn ip hops sip
+  assumes "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+      and "1 \<le> sn"
+      and *: "ip\<in>kD(rt (\<sigma> sip)) \<and> sn \<le> nsqn (rt (\<sigma> sip)) ip
+                                \<and> (nsqn (rt (\<sigma> sip)) ip = sn
+                                    \<longrightarrow> (the (dhops (rt (\<sigma> sip)) ip) \<le> hops
+                                          \<or> the (flag (rt (\<sigma> sip)) ip) = inv))"
+    shows "ip\<in>kD(rt (\<sigma>' sip)) \<and> sn \<le> nsqn (rt (\<sigma>' sip)) ip
+                              \<and> (nsqn (rt (\<sigma>' sip)) ip = sn
+                                  \<longrightarrow> (the (dhops (rt (\<sigma>' sip)) ip) \<le> hops
+                                        \<or> the (flag (rt (\<sigma>' sip)) ip) = inv))"
+  proof -
+    from assms(1) have "quality_increases (\<sigma> sip) (\<sigma>' sip)" ..
+    thus ?thesis using assms(2-3) by (rule quality_increases_rreq_rrep_props)
+  qed
+
+lemma rteq_quality_increases:
+  assumes "\<forall>j. j \<noteq> i \<longrightarrow> quality_increases (\<sigma> j) (\<sigma>' j)"
+      and "rt (\<sigma>' i) = rt (\<sigma> i)"
+    shows "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+  using assms by clarsimp (metis order_refl quality_increasesI rt_fresher_refl)
+
+definition msg_fresh :: "(ip \<Rightarrow> state) \<Rightarrow> msg \<Rightarrow> bool"
+where "msg_fresh \<sigma> m \<equiv>
+         case m of Rreq hopsc _ _ _ _ oipc osnc sipc \<Rightarrow> osnc \<ge> 1 \<and> (sipc \<noteq> oipc \<longrightarrow>
+                       oipc\<in>kD(rt (\<sigma> sipc)) \<and> nsqn (rt (\<sigma> sipc)) oipc \<ge> osnc
+                       \<and> (nsqn (rt (\<sigma> sipc)) oipc = osnc
+                             \<longrightarrow> (hopsc \<ge> the (dhops (rt (\<sigma> sipc)) oipc)
+                                  \<or> the (flag (rt (\<sigma> sipc)) oipc) = inv)))
+                   | Rrep hopsc dipc dsnc _ sipc \<Rightarrow> dsnc \<ge> 1 \<and> (sipc \<noteq> dipc \<longrightarrow>
+                       dipc\<in>kD(rt (\<sigma> sipc)) \<and> nsqn (rt (\<sigma> sipc)) dipc \<ge> dsnc
+                       \<and> (nsqn (rt (\<sigma> sipc)) dipc = dsnc
+                             \<longrightarrow> (hopsc \<ge> the (dhops (rt (\<sigma> sipc)) dipc)
+                                   \<or> the (flag (rt (\<sigma> sipc)) dipc) = inv)))
+                   | Rerr destsc sipc \<Rightarrow> (\<forall>ripc\<in>dom(destsc). (ripc\<in>kD(rt (\<sigma> sipc))
+                                         \<and> the (destsc ripc) - 1 \<le> nsqn (rt (\<sigma> sipc)) ripc))
+                   | _ \<Rightarrow> True"
+
+lemma msg_fresh [simp]:
+  "\<And>hops rreqid dip dsn dsk oip osn sip.
+           msg_fresh \<sigma> (Rreq hops rreqid dip dsn dsk oip osn sip) =
+                            (osn \<ge> 1 \<and> (sip \<noteq> oip \<longrightarrow> oip\<in>kD(rt (\<sigma> sip))
+                                     \<and> nsqn (rt (\<sigma> sip)) oip \<ge> osn
+                                     \<and> (nsqn (rt (\<sigma> sip)) oip = osn
+                                           \<longrightarrow> (hops \<ge> the (dhops (rt (\<sigma> sip)) oip)
+                                                \<or> the (flag (rt (\<sigma> sip)) oip) = inv))))"
+  "\<And>hops dip dsn oip sip. msg_fresh \<sigma> (Rrep hops dip dsn oip sip) =
+                            (dsn \<ge> 1 \<and> (sip \<noteq> dip \<longrightarrow> dip\<in>kD(rt (\<sigma> sip))
+                                     \<and> nsqn (rt (\<sigma> sip)) dip \<ge> dsn
+                                     \<and> (nsqn (rt (\<sigma> sip)) dip = dsn
+                                           \<longrightarrow> (hops \<ge> the (dhops (rt (\<sigma> sip)) dip))
+                                                 \<or> the (flag (rt (\<sigma> sip)) dip) = inv)))"
+  "\<And>dests sip.            msg_fresh \<sigma> (Rerr dests sip) =
+                            (\<forall>ripc\<in>dom(dests). (ripc\<in>kD(rt (\<sigma> sip))
+                                     \<and> the (dests ripc) - 1 \<le> nsqn (rt (\<sigma> sip)) ripc))"
+  "\<And>d dip.                msg_fresh \<sigma> (Newpkt d dip)   = True"
+  "\<And>d dip sip.            msg_fresh \<sigma> (Pkt d dip sip)  = True"
+  unfolding msg_fresh_def by simp_all
+
+lemma msg_fresh_inc_sn [simp, elim]:
+  "msg_fresh \<sigma> m \<Longrightarrow> rreq_rrep_sn m"
+  by (cases m) simp_all
+
+lemma recv_msg_fresh_inc_sn [simp, elim]:
+  "orecvmsg (msg_fresh) \<sigma> m \<Longrightarrow> recvmsg rreq_rrep_sn m"
+  by (cases m) simp_all
+
+lemma rreq_nsqn_is_fresh [simp]:
+  fixes \<sigma> msg hops rreqid dip dsn dsk oip osn sip
+  assumes "rreq_rrep_fresh (rt (\<sigma> sip)) (Rreq hops rreqid dip dsn dsk oip osn sip)"
+      and "rreq_rrep_sn (Rreq hops rreqid dip dsn dsk oip osn sip)"
+  shows "msg_fresh \<sigma> (Rreq hops rreqid dip dsn dsk oip osn sip)"
+        (is "msg_fresh \<sigma> ?msg")
+  proof -
+    let ?rt = "rt (\<sigma> sip)"
+    from assms(2) have "1 \<le> osn" by simp
+    thus ?thesis
+    unfolding msg_fresh_def
+    proof (simp only: msg.case, intro conjI impI)
+      assume "sip \<noteq> oip"
+      with assms(1) show "oip \<in> kD(?rt)" by simp
+    next
+      assume "sip \<noteq> oip"
+         and "nsqn ?rt oip = osn"
+      show "the (dhops ?rt oip) \<le> hops \<or> the (flag ?rt oip) = inv"
+      proof (cases "oip\<in>vD(?rt)")
+        assume "oip\<in>vD(?rt)"
+        hence "nsqn ?rt oip = sqn ?rt oip" ..
+        with \<open>nsqn ?rt oip = osn\<close> have "sqn ?rt oip = osn" by simp
+        with assms(1) and \<open>sip \<noteq> oip\<close> have "the (dhops ?rt oip) \<le> hops"
+          by simp
+        thus ?thesis ..
+      next
+        assume "oip\<notin>vD(?rt)"
+        moreover from assms(1) and \<open>sip \<noteq> oip\<close> have "oip\<in>kD(?rt)" by simp
+        ultimately have "oip\<in>iD(?rt)" by auto
+        hence "the (flag ?rt oip) = inv" ..
+        thus ?thesis ..
+      qed
+    next
+      assume "sip \<noteq> oip"
+      with assms(1) have "osn \<le> sqn ?rt oip" by auto
+      thus "osn \<le> nsqn (rt (\<sigma> sip)) oip"
+      proof (rule nat_le_eq_or_lt)
+        assume "osn < sqn ?rt oip"
+        hence "osn \<le> sqn ?rt oip - 1" by simp
+        also have "... \<le> nsqn ?rt oip" by (rule sqn_nsqn)
+        finally show "osn \<le> nsqn ?rt oip" .
+      next
+        assume "osn = sqn ?rt oip"
+        with assms(1) and \<open>sip \<noteq> oip\<close> have "oip\<in>kD(?rt)"
+                                       and "the (flag ?rt oip) = val"
+          by auto
+        hence "nsqn ?rt oip = sqn ?rt oip" ..
+        with \<open>osn = sqn ?rt oip\<close> have "nsqn ?rt oip = osn" by simp
+        thus "osn \<le> nsqn ?rt oip" by simp
+      qed
+    qed simp
+  qed
+
+lemma rrep_nsqn_is_fresh [simp]:
+  fixes \<sigma> msg hops dip dsn oip sip
+  assumes "rreq_rrep_fresh (rt (\<sigma> sip)) (Rrep hops dip dsn oip sip)"
+      and "rreq_rrep_sn (Rrep hops dip dsn oip sip)"
+  shows "msg_fresh \<sigma> (Rrep hops dip dsn oip sip)"
+        (is "msg_fresh \<sigma> ?msg")
+  proof -
+    let ?rt = "rt (\<sigma> sip)"
+    from assms have "sip \<noteq> dip \<longrightarrow> dip\<in>kD(?rt) \<and> sqn ?rt dip = dsn \<and> the (flag ?rt dip) = val"
+      by simp
+    hence "sip \<noteq> dip \<longrightarrow> dip\<in>kD(?rt) \<and> nsqn ?rt dip \<ge> dsn"
+    by clarsimp
+    with assms show "msg_fresh \<sigma> ?msg"
+      by clarsimp
+  qed
+
+lemma rerr_nsqn_is_fresh [simp]:
+  fixes \<sigma> msg dests sip
+  assumes "rerr_invalid (rt (\<sigma> sip)) (Rerr dests sip)"
+  shows "msg_fresh \<sigma> (Rerr dests sip)"
+        (is "msg_fresh \<sigma> ?msg")
+  proof -
+    let ?rt = "rt (\<sigma> sip)"
+    from assms have *: "(\<forall>rip\<in>dom(dests). (rip\<in>iD(rt (\<sigma> sip))
+                                     \<and> the (dests rip) = sqn (rt (\<sigma> sip)) rip))"
+      by clarsimp
+    have "(\<forall>rip\<in>dom(dests). (rip\<in>kD(rt (\<sigma> sip))
+                                     \<and> the (dests rip) - 1 \<le> nsqn (rt (\<sigma> sip)) rip))"
+    proof
+      fix rip
+      assume "rip \<in> dom dests"
+      with * have "rip\<in>iD(rt (\<sigma> sip))" and "the (dests rip) = sqn (rt (\<sigma> sip)) rip"
+        by auto
+
+      from this(2) have "the (dests rip) - 1 = sqn (rt (\<sigma> sip)) rip - 1" by simp
+      also have "... \<le> nsqn (rt (\<sigma> sip)) rip" by (rule sqn_nsqn)
+      finally have "the (dests rip) - 1 \<le> nsqn (rt (\<sigma> sip)) rip" .
+
+      with \<open>rip\<in>iD(rt (\<sigma> sip))\<close>
+        show "rip\<in>kD(rt (\<sigma> sip)) \<and> the (dests rip) - 1 \<le> nsqn (rt (\<sigma> sip)) rip"
+          by clarsimp
+    qed
+    thus "msg_fresh \<sigma> ?msg"
+      by simp
+  qed
+
+lemma quality_increases_msg_fresh [elim]:
+  assumes qinc: "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+      and "msg_fresh \<sigma> m"
+    shows "msg_fresh \<sigma>' m"
+  using assms(2)
+  proof (cases m)
+    fix hops rreqid dip dsn dsk oip osn sip
+    assume [simp]: "m = Rreq hops rreqid dip dsn dsk oip osn sip"
+       and "msg_fresh \<sigma> m"
+    then have "osn \<ge> 1" and "sip = oip \<or> (oip\<in>kD(rt (\<sigma> sip)) \<and> osn \<le> nsqn (rt (\<sigma> sip)) oip
+                                           \<and> (nsqn (rt (\<sigma> sip)) oip = osn
+                                                 \<longrightarrow> (the (dhops (rt (\<sigma> sip)) oip) \<le> hops
+                                                      \<or> the (flag (rt (\<sigma> sip)) oip) = inv)))"
+      by auto
+    from this(2) show ?thesis
+    proof
+      assume "sip = oip" with \<open>osn \<ge> 1\<close> show ?thesis by simp
+    next
+      assume "oip\<in>kD(rt (\<sigma> sip)) \<and> osn \<le> nsqn (rt (\<sigma> sip)) oip
+                                  \<and> (nsqn (rt (\<sigma> sip)) oip = osn
+                                      \<longrightarrow> (the (dhops (rt (\<sigma> sip)) oip) \<le> hops
+                                           \<or> the (flag (rt (\<sigma> sip)) oip) = inv))"
+      moreover from qinc have "quality_increases (\<sigma> sip) (\<sigma>' sip)" ..
+      ultimately have "oip\<in>kD(rt (\<sigma>' sip)) \<and> osn \<le> nsqn (rt (\<sigma>' sip)) oip
+                                           \<and> (nsqn (rt (\<sigma>' sip)) oip = osn
+                                              \<longrightarrow> (the (dhops (rt (\<sigma>' sip)) oip) \<le> hops
+                                                    \<or> the (flag (rt (\<sigma>' sip)) oip) = inv))"
+       using \<open>osn \<ge> 1\<close> by (rule quality_increases_rreq_rrep_props [rotated 2])
+      with \<open>osn \<ge> 1\<close> show "msg_fresh \<sigma>' m"
+        by (clarsimp)
+    qed
+  next
+    fix hops dip dsn oip sip
+    assume [simp]: "m = Rrep hops dip dsn oip sip"
+       and "msg_fresh \<sigma> m"
+    then have "dsn \<ge> 1" and "sip = dip \<or> (dip\<in>kD(rt (\<sigma> sip)) \<and> dsn \<le> nsqn (rt (\<sigma> sip)) dip
+                                           \<and> (nsqn (rt (\<sigma> sip)) dip = dsn
+                                                 \<longrightarrow> (the (dhops (rt (\<sigma> sip)) dip) \<le> hops
+                                                      \<or> the (flag (rt (\<sigma> sip)) dip) = inv)))"
+      by auto
+    from this(2) show "?thesis"
+    proof
+      assume "sip = dip" with \<open>dsn \<ge> 1\<close> show ?thesis by simp
+    next
+      assume "dip\<in>kD(rt (\<sigma> sip)) \<and> dsn \<le> nsqn (rt (\<sigma> sip)) dip
+                                  \<and> (nsqn (rt (\<sigma> sip)) dip = dsn
+                                      \<longrightarrow> (the (dhops (rt (\<sigma> sip)) dip) \<le> hops
+                                           \<or> the (flag (rt (\<sigma> sip)) dip) = inv))"
+      moreover from qinc have "quality_increases (\<sigma> sip) (\<sigma>' sip)" ..
+      ultimately have "dip\<in>kD(rt (\<sigma>' sip)) \<and> dsn \<le> nsqn (rt (\<sigma>' sip)) dip
+                                           \<and> (nsqn (rt (\<sigma>' sip)) dip = dsn
+                                              \<longrightarrow> (the (dhops (rt (\<sigma>' sip)) dip) \<le> hops
+                                                    \<or> the (flag (rt (\<sigma>' sip)) dip) = inv))"
+        using \<open>dsn \<ge> 1\<close> by (rule quality_increases_rreq_rrep_props [rotated 2])
+      with \<open>dsn \<ge> 1\<close> show "msg_fresh \<sigma>' m"
+        by clarsimp
+    qed
+  next
+    fix dests sip
+    assume [simp]: "m = Rerr dests sip"
+       and "msg_fresh \<sigma> m"
+    then have *: "\<forall>rip\<in>dom(dests). rip\<in>kD(rt (\<sigma> sip))
+                              \<and> the (dests rip) - 1 \<le> nsqn (rt (\<sigma> sip)) rip"
+      by simp
+    have "\<forall>rip\<in>dom(dests). rip\<in>kD(rt (\<sigma>' sip))
+                         \<and> the (dests rip) - 1 \<le> nsqn (rt (\<sigma>' sip)) rip"
+      proof
+        fix rip
+        assume "rip\<in>dom(dests)"
+        with * have "rip\<in>kD(rt (\<sigma> sip))" and "the (dests rip) - 1 \<le> nsqn (rt (\<sigma> sip)) rip"
+          by - (drule(1) bspec, clarsimp)+
+        moreover from qinc have "quality_increases (\<sigma> sip) (\<sigma>' sip)" by simp
+        ultimately show "rip\<in>kD(rt (\<sigma>' sip)) \<and> the (dests rip) - 1 \<le> nsqn (rt (\<sigma>' sip)) rip" ..
+      qed
+    thus ?thesis by simp
+  qed simp_all
+
+end

formal/afp/AODV/Seq_Invariants.thy

@@ -0,0 +1,643 @@
+(*  Title:       Seq_Invariants.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "Invariant proofs on individual processes"
+
+theory Seq_Invariants
+imports AWN.Invariants Aodv Aodv_Data Aodv_Predicates Fresher
+
+begin
+
+text \<open>
+  The proposition numbers are taken from the December 2013 version of
+  the Fehnker et al technical report.
+\<close>
+
+text \<open>Proposition 7.2\<close>
+
+lemma sequence_number_increases:
+  "paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). sn \<xi> \<le> sn \<xi>')"
+  by inv_cterms
+
+lemma sequence_number_one_or_bigger:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, _). 1 \<le> sn \<xi>)"
+  by (rule onll_step_to_invariantI [OF sequence_number_increases])
+     (auto simp: \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def)
+
+text \<open>We can get rid of the onl/onll if desired...\<close>
+
+lemma sequence_number_increases':
+  "paodv i \<TTurnstile>\<^sub>A (\<lambda>((\<xi>, _), _, (\<xi>', _)). sn \<xi> \<le> sn \<xi>')"
+  by (rule step_invariant_weakenE [OF sequence_number_increases]) (auto dest!: onllD)
+
+lemma sequence_number_one_or_bigger':
+  "paodv i \<TTurnstile> (\<lambda>(\<xi>, _). 1 \<le> sn \<xi>)"
+  by (rule invariant_weakenE [OF sequence_number_one_or_bigger]) auto
+
+lemma sip_in_kD:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). l \<in> ({PAodv-:7} \<union> {PAodv-:5} \<union> {PRrep-:0..PRrep-:1}
+                                     \<union> {PRreq-:0..PRreq-:3}) \<longrightarrow> sip \<xi> \<in> kD (rt \<xi>))"
+  by inv_cterms
+
+lemma rrep_1_update_changes:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). (l = PRrep-:1 \<longrightarrow>
+                        rt \<xi> \<noteq> update (rt \<xi>) (dip \<xi>) (dsn \<xi>, kno, val, hops \<xi> + 1, sip \<xi>, {})))"
+  by inv_cterms
+
+lemma addpreRT_partly_welldefined:
+  "paodv i \<TTurnstile>
+     onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). (l \<in> {PRreq-:16..PRreq-:18} \<union> {PRrep-:2..PRrep-:6} \<longrightarrow> dip \<xi> \<in> kD (rt \<xi>))
+                      \<and> (l \<in> {PRreq-:3..PRreq-:17} \<longrightarrow> oip \<xi> \<in> kD (rt \<xi>)))"
+  by inv_cterms
+
+text \<open>Proposition 7.38\<close>
+
+lemma includes_nhip:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). \<forall>dip\<in>kD(rt \<xi>). the (nhop (rt \<xi>) dip)\<in>kD(rt \<xi>))"
+  proof -
+    { fix ip and \<xi> \<xi>' :: state
+      assume "\<forall>dip\<in>kD (rt \<xi>). the (nhop (rt \<xi>) dip) \<in> kD (rt \<xi>)"
+         and "\<xi>' = \<xi>\<lparr>rt := update (rt \<xi>) ip (0, unk, val, Suc 0, ip, {})\<rparr>"
+      hence "\<forall>dip\<in>kD (rt \<xi>).
+               the (nhop (update (rt \<xi>) ip (0, unk, val, Suc 0, ip, {})) dip) = ip
+             \<or> the (nhop (update (rt \<xi>) ip (0, unk, val, Suc 0, ip, {})) dip) \<in> kD (rt \<xi>)"
+        by clarsimp (metis nhop_update_unk_val update_another)
+    } note one_hop = this
+    {  fix ip sip sn hops and \<xi> \<xi>' :: state
+       assume "\<forall>dip\<in>kD (rt \<xi>). the (nhop (rt \<xi>) dip) \<in> kD (rt \<xi>)"
+          and "\<xi>' = \<xi>\<lparr>rt := update (rt \<xi>) ip (sn, kno, val, Suc hops, sip, {})\<rparr>"
+          and "sip \<in> kD (rt \<xi>)"
+       hence "(the (nhop (update (rt \<xi>) ip (sn, kno, val, Suc hops, sip, {})) ip) = ip
+                 \<or> the (nhop (update (rt \<xi>) ip (sn, kno, val, Suc hops, sip, {})) ip) \<in> kD (rt \<xi>))
+               \<and> (\<forall>dip\<in>kD (rt \<xi>).
+                    the (nhop (update (rt \<xi>) ip (sn, kno, val, Suc hops, sip, {})) dip) = ip
+                    \<or> the (nhop (update (rt \<xi>) ip (sn, kno, val, Suc hops, sip, {})) dip) \<in> kD (rt \<xi>))"
+         by (metis kD_update_unchanged nhop_update_changed update_another)
+    } note nhip_is_sip = this
+    show ?thesis
+      by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf sip_in_kD]
+                              onl_invariant_sterms [OF aodv_wf addpreRT_partly_welldefined]
+                       solve: one_hop nhip_is_sip)
+  qed
+
+text \<open>Proposition 7.22: needed in Proposition 7.4\<close>
+
+lemma addpreRT_welldefined:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). (l \<in> {PRreq-:16..PRreq-:18} \<longrightarrow> dip \<xi> \<in> kD (rt \<xi>)) \<and>
+                               (l = PRreq-:17 \<longrightarrow> oip \<xi> \<in> kD (rt \<xi>)) \<and>                  
+                               (l = PRrep-:5  \<longrightarrow> dip \<xi> \<in> kD (rt \<xi>)) \<and>
+                               (l = PRrep-:6  \<longrightarrow> (the (nhop (rt \<xi>) (dip \<xi>))) \<in> kD (rt \<xi>)))"
+  (is "_ \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V ?P")
+  unfolding invariant_def
+  proof
+    fix s
+    assume "s \<in> reachable (paodv i) TT"
+    then obtain \<xi> p where "s = (\<xi>, p)"
+                      and "(\<xi>, p) \<in> reachable (paodv i) TT"
+      by (metis prod.exhaust)
+    have "onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V ?P (\<xi>, p)"
+    proof (rule onlI)
+      fix l
+      assume "l \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p"
+      with \<open>(\<xi>, p) \<in> reachable (paodv i) TT\<close>
+        have I1: "l \<in> {PRreq-:16..PRreq-:18} \<longrightarrow> dip \<xi> \<in> kD(rt \<xi>)"
+         and I2: "l = PRreq-:17 \<longrightarrow> oip \<xi> \<in> kD(rt \<xi>)"
+         and I3: "l \<in> {PRrep-:2..PRrep-:6}  \<longrightarrow> dip \<xi> \<in> kD(rt \<xi>)"
+         by (auto dest!: invariantD [OF addpreRT_partly_welldefined])
+      moreover from \<open>(\<xi>, p) \<in> reachable (paodv i) TT\<close> \<open>l \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p\<close> and I3
+        have "l = PRrep-:6  \<longrightarrow> (the (nhop (rt \<xi>) (dip \<xi>))) \<in> kD(rt \<xi>)"
+          by (auto dest!: invariantD [OF includes_nhip])
+      ultimately show "?P (\<xi>, l)"
+        by simp
+    qed
+    with \<open>s = (\<xi>, p)\<close> show "onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V ?P s"
+      by simp
+  qed
+
+text \<open>Proposition 7.4\<close>
+
+lemma known_destinations_increase:
+  "paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). kD (rt \<xi>) \<subseteq> kD (rt \<xi>'))"
+  by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf addpreRT_welldefined]
+                 simp add: subset_insertI)
+
+text \<open>Proposition 7.5\<close>
+
+lemma rreqs_increase:
+  "paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). rreqs \<xi> \<subseteq> rreqs \<xi>')"
+  by (inv_cterms simp add: subset_insertI)
+
+lemma dests_bigger_than_sqn:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). l \<in> {PAodv-:15..PAodv-:19}
+                                 \<union> {PPkt-:7..PPkt-:11}
+                                 \<union> {PRreq-:9..PRreq-:13}
+                                 \<union> {PRreq-:21..PRreq-:25}
+                                 \<union> {PRrep-:10..PRrep-:14}
+                                 \<union> {PRerr-:1..PRerr-:5}
+                         \<longrightarrow> (\<forall>ip\<in>dom(dests \<xi>). ip\<in>kD(rt \<xi>) \<and> sqn (rt \<xi>) ip \<le> the (dests \<xi> ip)))"
+  proof -
+    have sqninv:
+      "\<And>dests rt rsn ip.
+       \<lbrakk> \<forall>ip\<in>dom(dests). ip\<in>kD(rt) \<and> sqn rt ip \<le> the (dests ip); dests ip = Some rsn \<rbrakk>
+        \<Longrightarrow> sqn (invalidate rt dests) ip \<le> rsn"
+        by (rule sqn_invalidate_in_dests [THEN eq_imp_le], assumption) auto
+    have indests:
+      "\<And>dests rt rsn ip.
+       \<lbrakk> \<forall>ip\<in>dom(dests). ip\<in>kD(rt) \<and> sqn rt ip \<le> the (dests ip); dests ip = Some rsn \<rbrakk>
+        \<Longrightarrow> ip\<in>kD(rt) \<and> sqn rt ip \<le> rsn"
+        by (metis domI option.sel)
+    show ?thesis
+      by inv_cterms
+         (clarsimp split: if_split_asm option.split_asm
+                   elim!: sqninv indests)+
+  qed
+
+text \<open>Proposition 7.6\<close>
+
+lemma sqns_increase:
+   "paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). \<forall>ip. sqn (rt \<xi>) ip \<le> sqn (rt \<xi>') ip)"
+  proof -
+    { fix \<xi> :: state
+      assume *: "\<forall>ip\<in>dom(dests \<xi>). ip \<in> kD (rt \<xi>) \<and> sqn (rt \<xi>) ip \<le> the (dests \<xi> ip)"
+      have "\<forall>ip. sqn (rt \<xi>) ip \<le> sqn (invalidate (rt \<xi>) (dests \<xi>)) ip"
+      proof
+        fix ip
+        from * have "ip\<notin>dom(dests \<xi>) \<or> sqn (rt \<xi>) ip \<le> the (dests \<xi> ip)" by simp
+        thus "sqn (rt \<xi>) ip \<le> sqn (invalidate (rt \<xi>) (dests \<xi>)) ip"
+          by (metis domI invalidate_sqn option.sel)
+      qed
+    } note solve_invalidate = this
+    show ?thesis
+      by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf addpreRT_welldefined]
+                              onl_invariant_sterms [OF aodv_wf dests_bigger_than_sqn]
+                    simp add: solve_invalidate)
+  qed
+
+text \<open>Proposition 7.7\<close>
+
+lemma ip_constant:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, _). ip \<xi> = i)"
+  by (inv_cterms simp add: \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def)
+
+text \<open>Proposition 7.8\<close>
+
+lemma sender_ip_valid':
+  "paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), a, _). anycast (\<lambda>m. not_Pkt m \<longrightarrow> msg_sender m = ip \<xi>) a)"
+  by inv_cterms
+
+lemma sender_ip_valid:
+  "paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), a, _). anycast (\<lambda>m. not_Pkt m \<longrightarrow> msg_sender m = i) a)"
+  by (rule step_invariant_weaken_with_invariantE [OF ip_constant sender_ip_valid'])
+     (auto dest!: onlD onllD)
+
+lemma received_msg_inv:
+  "paodv i \<TTurnstile> (recvmsg P \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). l \<in> {PAodv-:1} \<longrightarrow> P (msg \<xi>))"
+  by inv_cterms
+
+text \<open>Proposition 7.9\<close>
+
+lemma sip_not_ip':
+  "paodv i \<TTurnstile> (recvmsg (\<lambda>m. not_Pkt m \<longrightarrow> msg_sender m \<noteq> i) \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, _). sip \<xi> \<noteq> ip \<xi>)"
+  by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf received_msg_inv]
+                          onl_invariant_sterms [OF aodv_wf ip_constant [THEN invariant_restrict_inD]]
+                simp add: clear_locals_sip_not_ip') clarsimp+
+
+lemma sip_not_ip:
+  "paodv i \<TTurnstile> (recvmsg (\<lambda>m. not_Pkt m \<longrightarrow> msg_sender m \<noteq> i) \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, _). sip \<xi> \<noteq> i)"
+  by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf received_msg_inv]
+                          onl_invariant_sterms [OF aodv_wf ip_constant [THEN invariant_restrict_inD]]
+                simp add: clear_locals_sip_not_ip') clarsimp+
+
+text \<open>Neither \<open>sip_not_ip'\<close> nor \<open>sip_not_ip\<close> is needed to show loop freedom.\<close>
+
+text \<open>Proposition 7.10\<close>
+
+lemma hop_count_positive:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, _). \<forall>ip\<in>kD (rt \<xi>). the (dhops (rt \<xi>) ip) \<ge> 1)"
+  by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf addpreRT_welldefined]) auto
+
+lemma rreq_dip_in_vD_dip_eq_ip:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). (l \<in> {PRreq-:16..PRreq-:18} \<longrightarrow> dip \<xi> \<in> vD(rt \<xi>))
+                            \<and> (l \<in> {PRreq-:5, PRreq-:6} \<longrightarrow> dip \<xi> = ip \<xi>)
+                            \<and> (l \<in> {PRreq-:15..PRreq-:18} \<longrightarrow> dip \<xi> \<noteq> ip \<xi>))"
+  proof (inv_cterms, elim conjE)
+    fix l \<xi> pp p'
+    assume "(\<xi>, pp) \<in> reachable (paodv i) TT"
+       and "{PRreq-:17}\<lbrakk>\<lambda>\<xi>. \<xi>\<lparr>rt := the (addpreRT (rt \<xi>) (oip \<xi>) {the (nhop (rt \<xi>) (dip \<xi>))})\<rparr>\<rbrakk> p'
+              \<in> sterms \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pp"
+       and "l = PRreq-:17"
+       and "dip \<xi> \<in> vD (rt \<xi>)"
+    from this(1-3) have "oip \<xi> \<in> kD (rt \<xi>)"
+      by (auto dest: onl_invariant_sterms [OF aodv_wf addpreRT_welldefined, where l="PRreq-:17"])
+    with \<open>dip \<xi> \<in> vD (rt \<xi>)\<close>
+      show "dip \<xi> \<in> vD (the (addpreRT (rt \<xi>) (oip \<xi>) {the (nhop (rt \<xi>) (dip \<xi>))}))" by simp
+  qed
+
+text \<open>Proposition 7.11\<close>
+
+lemma anycast_msg_zhops:
+  "\<And>rreqid dip dsn dsk oip osn sip.
+      paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(_, a, _). anycast msg_zhops a)"
+  proof (inv_cterms inv add:
+           onl_invariant_sterms [OF aodv_wf rreq_dip_in_vD_dip_eq_ip [THEN invariant_restrict_inD]]
+           onl_invariant_sterms [OF aodv_wf hop_count_positive [THEN invariant_restrict_inD]],
+         elim conjE)
+    fix l \<xi> a pp p' pp'
+    assume "(\<xi>, pp) \<in> reachable (paodv i) TT"
+       and "{PRreq-:18}unicast(\<lambda>\<xi>. the (nhop (rt \<xi>) (oip \<xi>)),
+               \<lambda>\<xi>. Rrep (the (dhops (rt \<xi>) (dip \<xi>))) (dip \<xi>) (sqn (rt \<xi>) (dip \<xi>)) (oip \<xi>) (ip \<xi>)).
+                     p' \<triangleright> pp' \<in> sterms \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pp"
+       and "l = PRreq-:18"
+       and "a = unicast (the (nhop (rt \<xi>) (oip \<xi>)))
+                 (Rrep (the (dhops (rt \<xi>) (dip \<xi>))) (dip \<xi>) (sqn (rt \<xi>) (dip \<xi>)) (oip \<xi>) (ip \<xi>))"
+       and *: "\<forall>ip\<in>kD (rt \<xi>). Suc 0 \<le> the (dhops (rt \<xi>) ip)"
+       and "dip \<xi> \<in> vD (rt \<xi>)"
+    from \<open>dip \<xi> \<in> vD (rt \<xi>)\<close> have "dip \<xi> \<in> kD (rt \<xi>)"
+      by (rule vD_iD_gives_kD(1))
+    with * have "Suc 0 \<le> the (dhops (rt \<xi>) (dip \<xi>))" ..
+    thus "0 < the (dhops (rt \<xi>) (dip \<xi>))" by simp
+  qed
+
+lemma hop_count_zero_oip_dip_sip:
+  "paodv i \<TTurnstile> (recvmsg msg_zhops \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                 (l\<in>{PAodv-:4..PAodv-:5} \<union> {PRreq-:n|n. True} \<longrightarrow>
+                          (hops \<xi> = 0 \<longrightarrow> oip \<xi> = sip \<xi>))
+                 \<and>
+                 ((l\<in>{PAodv-:6..PAodv-:7} \<union> {PRrep-:n|n. True} \<longrightarrow>
+                          (hops \<xi> = 0 \<longrightarrow> dip \<xi> = sip \<xi>))))"
+  by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf received_msg_inv]) auto
+
+lemma osn_rreq:
+  "paodv i \<TTurnstile> (recvmsg rreq_rrep_sn \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                                    l \<in> {PAodv-:4, PAodv-:5} \<union> {PRreq-:n|n. True} \<longrightarrow> 1 \<le> osn \<xi>)"
+  by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf received_msg_inv]) clarsimp
+
+lemma osn_rreq':
+  "paodv i \<TTurnstile> (recvmsg (\<lambda>m. rreq_rrep_sn m \<and> msg_zhops m) \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                                    l \<in> {PAodv-:4, PAodv-:5} \<union> {PRreq-:n|n. True} \<longrightarrow> 1 \<le> osn \<xi>)"
+  proof (rule invariant_weakenE [OF osn_rreq])
+    fix a
+    assume "recvmsg (\<lambda>m. rreq_rrep_sn m \<and> msg_zhops m) a"
+    thus "recvmsg rreq_rrep_sn a"
+      by (cases a) simp_all
+  qed
+
+lemma dsn_rrep:
+  "paodv i \<TTurnstile> (recvmsg rreq_rrep_sn \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                                    l \<in> {PAodv-:6, PAodv-:7} \<union> {PRrep-:n|n. True} \<longrightarrow> 1 \<le> dsn \<xi>)"
+  by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf received_msg_inv]) clarsimp
+
+lemma dsn_rrep':
+  "paodv i \<TTurnstile> (recvmsg (\<lambda>m. rreq_rrep_sn m \<and> msg_zhops m) \<rightarrow>)  onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                                    l \<in> {PAodv-:6, PAodv-:7} \<union> {PRrep-:n|n. True} \<longrightarrow> 1 \<le> dsn \<xi>)"
+  proof (rule invariant_weakenE [OF dsn_rrep])
+    fix a
+    assume "recvmsg (\<lambda>m. rreq_rrep_sn m \<and> msg_zhops m) a"
+    thus "recvmsg rreq_rrep_sn a"
+      by (cases a) simp_all
+  qed
+
+lemma hop_count_zero_oip_dip_sip':
+  "paodv i \<TTurnstile> (recvmsg (\<lambda>m. rreq_rrep_sn m \<and> msg_zhops m) \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                 (l\<in>{PAodv-:4..PAodv-:5} \<union> {PRreq-:n|n. True} \<longrightarrow>
+                          (hops \<xi> = 0 \<longrightarrow> oip \<xi> = sip \<xi>))
+                 \<and>
+                 ((l\<in>{PAodv-:6..PAodv-:7} \<union> {PRrep-:n|n. True} \<longrightarrow>
+                          (hops \<xi> = 0 \<longrightarrow> dip \<xi> = sip \<xi>))))"
+  proof (rule invariant_weakenE [OF hop_count_zero_oip_dip_sip])
+    fix a
+    assume "recvmsg (\<lambda>m. rreq_rrep_sn m \<and> msg_zhops m) a"
+    thus "recvmsg msg_zhops a"
+      by (cases a) simp_all
+  qed
+
+text \<open>Proposition 7.12\<close>
+
+lemma zero_seq_unk_hops_one':
+  "paodv i \<TTurnstile> (recvmsg (\<lambda>m. rreq_rrep_sn m \<and> msg_zhops m) \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, _).
+                 \<forall>dip\<in>kD(rt \<xi>). (sqn (rt \<xi>) dip = 0 \<longrightarrow> sqnf (rt \<xi>) dip = unk)
+                              \<and> (sqnf (rt \<xi>) dip = unk \<longrightarrow> the (dhops (rt \<xi>) dip) = 1)
+                              \<and> (the (dhops (rt \<xi>) dip) = 1 \<longrightarrow> the (nhop (rt \<xi>) dip) = dip))"
+  proof -
+  { fix dip and \<xi> :: state and P
+    assume "sqn (invalidate (rt \<xi>) (dests \<xi>)) dip = 0"
+       and all: "\<forall>ip. sqn (rt \<xi>) ip \<le> sqn (invalidate (rt \<xi>) (dests \<xi>)) ip"
+       and *: "sqn (rt \<xi>) dip = 0 \<Longrightarrow> P \<xi> dip"
+    have "P \<xi> dip"
+    proof -
+      from all have "sqn (rt \<xi>) dip \<le> sqn (invalidate (rt \<xi>) (dests \<xi>)) dip" ..
+      with \<open>sqn (invalidate (rt \<xi>) (dests \<xi>)) dip = 0\<close> have "sqn (rt \<xi>) dip = 0" by simp
+      thus "P \<xi> dip" by (rule *)
+    qed
+  } note sqn_invalidate_zero [elim!] = this
+
+  { fix dsn hops :: nat and sip oip rt and ip dip :: ip
+    assume "\<forall>dip\<in>kD(rt).
+                (sqn rt dip = 0 \<longrightarrow> \<pi>\<^sub>3(the (rt dip)) = unk) \<and>
+                (\<pi>\<^sub>3(the (rt dip)) = unk \<longrightarrow> the (dhops rt dip) = Suc 0) \<and>
+                (the (dhops rt dip) = Suc 0 \<longrightarrow> the (nhop rt dip) = dip)"
+       and "hops = 0 \<longrightarrow> sip = dip"
+       and "Suc 0 \<le> dsn"
+       and "ip \<noteq> dip \<longrightarrow> ip\<in>kD(rt)"
+    hence "the (dhops (update rt dip (dsn, kno, val, Suc hops, sip, {})) ip) = Suc 0 \<longrightarrow>
+           the (nhop (update rt dip (dsn, kno, val, Suc hops, sip, {})) ip) = ip"
+      by - (rule update_cases, auto simp add: sqn_def dest!: bspec)
+  } note prreq_ok1 [simp] = this
+
+  { fix ip dsn hops sip oip rt dip
+    assume "\<forall>dip\<in>kD(rt).
+                (sqn rt dip = 0 \<longrightarrow> \<pi>\<^sub>3(the (rt dip)) = unk) \<and>
+                (\<pi>\<^sub>3(the (rt dip)) = unk \<longrightarrow> the (dhops rt dip) = Suc 0) \<and>
+                (the (dhops rt dip) = Suc 0 \<longrightarrow> the (nhop rt dip) = dip)"
+       and "Suc 0 \<le> dsn"
+       and "ip \<noteq> dip \<longrightarrow> ip\<in>kD(rt)"
+    hence "\<pi>\<^sub>3(the (update rt dip (dsn, kno, val, Suc hops, sip, {}) ip)) = unk \<longrightarrow>
+           the (dhops (update rt dip (dsn, kno, val, Suc hops, sip, {})) ip) = Suc 0"
+      by - (rule update_cases, auto simp add: sqn_def sqnf_def dest!: bspec)
+  } note prreq_ok2 [simp] = this
+
+  { fix ip dsn hops sip oip rt dip
+    assume "\<forall>dip\<in>kD(rt).
+                (sqn rt dip = 0 \<longrightarrow> \<pi>\<^sub>3(the (rt dip)) = unk) \<and>
+                (\<pi>\<^sub>3(the (rt dip)) = unk \<longrightarrow> the (dhops rt dip) = Suc 0) \<and>
+                (the (dhops rt dip) = Suc 0 \<longrightarrow> the (nhop rt dip) = dip)"
+       and "Suc 0 \<le> dsn"
+       and "ip \<noteq> dip \<longrightarrow> ip\<in>kD(rt)"
+    hence "sqn (update rt dip (dsn, kno, val, Suc hops, sip, {})) ip = 0 \<longrightarrow>
+           \<pi>\<^sub>3 (the (update rt dip (dsn, kno, val, Suc hops, sip, {}) ip)) = unk"
+      by - (rule update_cases, auto simp add: sqn_def dest!: bspec)
+  } note prreq_ok3 [simp] = this
+
+  { fix rt sip
+    assume "\<forall>dip\<in>kD rt.
+              (sqn rt dip = 0 \<longrightarrow> \<pi>\<^sub>3(the (rt dip)) = unk) \<and>
+              (\<pi>\<^sub>3(the (rt dip)) = unk \<longrightarrow> the (dhops rt dip) = Suc 0) \<and>
+              (the (dhops rt dip) = Suc 0 \<longrightarrow> the (nhop rt dip) = dip)"
+    hence "\<forall>dip\<in>kD rt.
+          (sqn (update rt sip (0, unk, val, Suc 0, sip, {})) dip = 0 \<longrightarrow>
+           \<pi>\<^sub>3(the (update rt sip (0, unk, val, Suc 0, sip, {}) dip)) = unk)
+        \<and> (\<pi>\<^sub>3(the (update rt sip (0, unk, val, Suc 0, sip, {}) dip)) = unk \<longrightarrow>
+           the (dhops (update rt sip (0, unk, val, Suc 0, sip, {})) dip) = Suc 0)
+        \<and> (the (dhops (update rt sip (0, unk, val, Suc 0, sip, {})) dip) = Suc 0 \<longrightarrow>
+           the (nhop (update rt sip (0, unk, val, Suc 0, sip, {})) dip) = dip)"
+    by - (rule update_cases, simp_all add: sqnf_def sqn_def)
+  } note prreq_ok4 [simp] = this
+
+  have prreq_ok5 [simp]: "\<And>sip rt.
+    \<pi>\<^sub>3(the (update rt sip (0, unk, val, Suc 0, sip, {}) sip)) = unk \<longrightarrow>
+    the (dhops (update rt sip (0, unk, val, Suc 0, sip, {})) sip) = Suc 0"
+    by (rule update_cases) simp_all
+
+  have prreq_ok6 [simp]: "\<And>sip rt.
+    sqn (update rt sip (0, unk, val, Suc 0, sip, {})) sip = 0 \<longrightarrow>
+    \<pi>\<^sub>3 (the (update rt sip (0, unk, val, Suc 0, sip, {}) sip)) = unk"
+    by (rule update_cases) simp_all
+
+  show ?thesis
+    by (inv_cterms inv add: onl_invariant_sterms_TT [OF aodv_wf addpreRT_welldefined]
+                            onl_invariant_sterms [OF aodv_wf hop_count_zero_oip_dip_sip']
+                            seq_step_invariant_sterms_TT [OF sqns_increase aodv_wf aodv_trans]
+                            onl_invariant_sterms [OF aodv_wf osn_rreq']
+                            onl_invariant_sterms [OF aodv_wf dsn_rrep']) clarsimp+
+  qed
+
+lemma zero_seq_unk_hops_one:
+  "paodv i \<TTurnstile> (recvmsg (\<lambda>m. rreq_rrep_sn m \<and> msg_zhops m) \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, _).
+                 \<forall>dip\<in>kD(rt \<xi>). (sqn (rt \<xi>) dip = 0 \<longrightarrow> (sqnf (rt \<xi>) dip = unk
+                                                         \<and> the (dhops (rt \<xi>) dip) = 1
+                                                         \<and> the (nhop (rt \<xi>) dip) = dip)))"
+  by (rule invariant_weakenE [OF zero_seq_unk_hops_one']) auto
+
+lemma kD_unk_or_atleast_one:
+  "paodv i \<TTurnstile> (recvmsg rreq_rrep_sn \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                               \<forall>dip\<in>kD(rt \<xi>). \<pi>\<^sub>3(the (rt \<xi> dip)) = unk \<or> 1 \<le> \<pi>\<^sub>2(the (rt \<xi> dip)))"
+  proof -
+    { fix sip rt dsn1 dsn2 dsk1 dsk2 flag1 flag2 hops1 hops2 nhip1 nhip2 pre1 pre2
+      assume "dsk1 = unk \<or> Suc 0 \<le> dsn2"
+      hence "\<pi>\<^sub>3(the (update rt sip (dsn1, dsk1, flag1, hops1, nhip1, pre1) sip)) = unk
+            \<or> Suc 0 \<le> sqn (update rt sip (dsn2, dsk2, flag2, hops2, nhip2, pre2)) sip"
+        unfolding update_def by (cases "dsk1 =unk") (clarsimp split: option.split)+
+    } note fromsip [simp] = this
+
+    { fix dip sip rt dsn1 dsn2 dsk1 dsk2 flag1 flag2 hops1 hops2 nhip1 nhip2 pre1 pre2
+      assume allkd: "\<forall>dip\<in>kD(rt). \<pi>\<^sub>3(the (rt dip)) = unk \<or> Suc 0 \<le> sqn rt dip"
+         and    **: "dsk1 = unk \<or> Suc 0 \<le> dsn2"
+      have "\<forall>dip\<in>kD(rt). \<pi>\<^sub>3(the (update rt sip (dsn1, dsk1, flag1, hops1, nhip1, pre1) dip)) = unk
+            \<or> Suc 0 \<le> sqn (update rt sip (dsn2, dsk2, flag2, hops2, nhip2, pre2)) dip"
+        (is "\<forall>dip\<in>kD(rt). ?prop dip")
+      proof
+        fix dip
+        assume "dip\<in>kD(rt)"
+        thus "?prop dip"
+        proof (cases "dip = sip")
+          assume "dip = sip"
+          with ** show ?thesis
+            by simp
+        next
+          assume "dip \<noteq> sip"
+          with \<open>dip\<in>kD(rt)\<close> allkd show ?thesis
+            by simp
+        qed
+      qed
+    } note solve_update [simp] = this
+
+    { fix dip rt dests
+      assume  *: "\<forall>ip\<in>dom(dests). ip\<in>kD(rt) \<and> sqn rt ip \<le> the (dests ip)"
+         and **: "\<forall>ip\<in>kD(rt). \<pi>\<^sub>3(the (rt ip)) = unk \<or> Suc 0 \<le> sqn rt ip"
+      have "\<forall>dip\<in>kD(rt). \<pi>\<^sub>3(the (rt dip)) = unk \<or> Suc 0 \<le> sqn (invalidate rt dests) dip"
+      proof
+        fix dip
+        assume "dip\<in>kD(rt)"
+        with ** have "\<pi>\<^sub>3(the (rt dip)) = unk \<or> Suc 0 \<le> sqn rt dip" ..
+        thus "\<pi>\<^sub>3 (the (rt dip)) = unk \<or> Suc 0 \<le> sqn (invalidate rt dests) dip"
+        proof
+          assume "\<pi>\<^sub>3(the (rt dip)) = unk" thus ?thesis ..
+        next
+          assume "Suc 0 \<le> sqn rt dip"
+          have "Suc 0 \<le> sqn (invalidate rt dests) dip"
+          proof (cases "dip\<in>dom(dests)")
+            assume "dip\<in>dom(dests)"
+            with * have "sqn rt dip \<le> the (dests dip)" by simp
+            with \<open>Suc 0 \<le> sqn rt dip\<close> have "Suc 0 \<le> the (dests dip)" by simp
+            with \<open>dip\<in>dom(dests)\<close> \<open>dip\<in>kD(rt)\<close> [THEN kD_Some] show ?thesis
+              unfolding invalidate_def sqn_def by auto
+          next
+            assume "dip\<notin>dom(dests)"
+            with \<open>Suc 0 \<le> sqn rt dip\<close> \<open>dip\<in>kD(rt)\<close> [THEN kD_Some] show ?thesis
+              unfolding invalidate_def sqn_def by auto
+          qed
+        thus ?thesis by (rule disjI2)
+        qed
+      qed
+    } note solve_invalidate [simp] = this
+
+    show ?thesis
+      by (inv_cterms inv add: onl_invariant_sterms_TT [OF aodv_wf addpreRT_welldefined]
+                              onl_invariant_sterms [OF aodv_wf dests_bigger_than_sqn
+                                                               [THEN invariant_restrict_inD]]
+                              onl_invariant_sterms [OF aodv_wf osn_rreq]
+                              onl_invariant_sterms [OF aodv_wf dsn_rrep]
+                    simp add: proj3_inv proj2_eq_sqn)
+  qed
+
+text \<open>Proposition 7.13\<close>
+
+lemma rreq_rrep_sn_any_step_invariant:
+  "paodv i \<TTurnstile>\<^sub>A (recvmsg rreq_rrep_sn \<rightarrow>) onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(_, a, _). anycast rreq_rrep_sn a)"
+  proof -
+    have sqnf_kno: "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                                      (l \<in> {PRreq-:16..PRreq-:18} \<longrightarrow> sqnf (rt \<xi>) (dip \<xi>) = kno))"
+      by (inv_cterms inv add: onl_invariant_sterms_TT [OF aodv_wf addpreRT_welldefined])
+    show ?thesis
+      by (inv_cterms inv add: onl_invariant_sterms_TT [OF aodv_wf addpreRT_welldefined]
+                              onl_invariant_sterms [OF aodv_wf sequence_number_one_or_bigger
+                                                               [THEN invariant_restrict_inD]]
+                              onl_invariant_sterms [OF aodv_wf kD_unk_or_atleast_one]
+                              onl_invariant_sterms_TT [OF aodv_wf sqnf_kno]
+                              onl_invariant_sterms [OF aodv_wf osn_rreq]
+                              onl_invariant_sterms [OF aodv_wf dsn_rrep])
+         (auto simp: proj2_eq_sqn)
+  qed
+
+text \<open>Proposition 7.14\<close>
+
+lemma rreq_rrep_fresh_any_step_invariant:
+  "paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), a, _). anycast (rreq_rrep_fresh (rt \<xi>)) a)"
+  proof -                                                      
+    have rreq_oip: "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                       (l \<in> {PRreq-:3, PRreq-:4, PRreq-:15, PRreq-:27}
+                               \<longrightarrow> oip \<xi> \<in> kD(rt \<xi>)
+                                 \<and> (sqn (rt \<xi>) (oip \<xi>) > (osn \<xi>)
+                                     \<or> (sqn (rt \<xi>) (oip \<xi>) = (osn \<xi>)
+                                        \<and> the (dhops (rt \<xi>) (oip \<xi>)) \<le> Suc (hops \<xi>)
+                                        \<and> the (flag (rt \<xi>) (oip \<xi>)) = val))))"
+      proof inv_cterms
+        fix l \<xi> l' pp p'
+        assume "(\<xi>, pp) \<in> reachable (paodv i) TT"
+           and "{PRreq-:2}\<lbrakk>\<lambda>\<xi>. \<xi>\<lparr>rt :=
+                update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})\<rparr>\<rbrakk> p' \<in> sterms \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pp"
+           and "l' = PRreq-:3"
+        show "osn \<xi> < sqn (update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})) (oip \<xi>)
+           \<or> (sqn (update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})) (oip \<xi>) = osn \<xi>
+             \<and> the (dhops (update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})) (oip \<xi>))
+                                                                            \<le> Suc (hops \<xi>)
+             \<and> the (flag (update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})) (oip \<xi>))
+                                                                            = val)"
+          unfolding update_def by (clarsimp split: option.split)
+                                  (metis linorder_neqE_nat not_less)
+      qed
+
+    have rrep_prrep: "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+          (l \<in> {PRrep-:2..PRrep-:7} \<longrightarrow> (dip \<xi> \<in> kD(rt \<xi>)
+                                        \<and> sqn (rt \<xi>) (dip \<xi>) = dsn \<xi>
+                                        \<and> the (dhops (rt \<xi>) (dip \<xi>)) = Suc (hops \<xi>)
+                                        \<and> the (flag (rt \<xi>) (dip \<xi>)) = val
+                                        \<and> the (nhop (rt \<xi>) (dip \<xi>)) \<in> kD (rt \<xi>))))"
+      by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf rrep_1_update_changes]
+                              onl_invariant_sterms [OF aodv_wf sip_in_kD])
+
+    show ?thesis
+      by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf rreq_oip]
+                              onl_invariant_sterms [OF aodv_wf rreq_dip_in_vD_dip_eq_ip]
+                              onl_invariant_sterms [OF aodv_wf rrep_prrep])
+  qed
+
+text \<open>Proposition 7.15\<close>
+
+lemma rerr_invalid_any_step_invariant:
+  "paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), a, _). anycast (rerr_invalid (rt \<xi>)) a)"
+  proof -
+    have dests_inv: "paodv i \<TTurnstile>
+                      onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). (l \<in> {PAodv-:15, PPkt-:7, PRreq-:9,
+                                            PRreq-:21, PRrep-:10, PRerr-:1}
+                          \<longrightarrow> (\<forall>ip\<in>dom(dests \<xi>). ip\<in>vD(rt \<xi>)))
+                         \<and> (l \<in> {PAodv-:16..PAodv-:19}
+                              \<union> {PPkt-:8..PPkt-:11}
+                              \<union> {PRreq-:10..PRreq-:13}
+                              \<union> {PRreq-:22..PRreq-:25}
+                              \<union> {PRrep-:11..PRrep-:14}
+                              \<union> {PRerr-:2..PRerr-:5} \<longrightarrow> (\<forall>ip\<in>dom(dests \<xi>). ip\<in>iD(rt \<xi>)
+                                                          \<and> the (dests \<xi> ip) = sqn (rt \<xi>) ip))
+                         \<and> (l = PPkt-:14 \<longrightarrow> dip \<xi>\<in>iD(rt \<xi>)))"
+      by inv_cterms (clarsimp split: if_split_asm option.split_asm simp: domIff)+
+    show ?thesis
+      by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf dests_inv])
+  qed
+
+text \<open>Proposition 7.16\<close>
+
+text \<open>
+  Some well-definedness obligations are irrelevant for the Isabelle development:
+
+  \begin{enumerate}
+  \item In each routing table there is at most one entry for each destination: guaranteed by type.
+
+  \item In each store of queued data packets there is at most one data queue for
+        each destination: guaranteed by structure.
+
+  \item Whenever a set of pairs @{term "(rip, rsn)"} is assigned to the variable
+        @{term "dests"} of type @{typ "ip \<rightharpoonup> sqn"}, or to the first argument of
+        the function @{term "rerr"}, this set is a partial function, i.e., there
+        is at most one entry @{term "(rip, rsn)"} for each destination
+        @{term "rip"}: guaranteed by type.
+  \end{enumerate}
+\<close>
+
+lemma dests_vD_inc_sqn:
+  "paodv i \<TTurnstile>
+        onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). (l \<in> {PAodv-:15, PPkt-:7, PRreq-:9, PRreq-:21, PRrep-:10}
+             \<longrightarrow> (\<forall>ip\<in>dom(dests \<xi>). ip\<in>vD(rt \<xi>) \<and> the (dests \<xi> ip) = inc (sqn (rt \<xi>) ip)))
+           \<and> (l = PRerr-:1
+             \<longrightarrow> (\<forall>ip\<in>dom(dests \<xi>). ip\<in>vD(rt \<xi>) \<and> the (dests \<xi> ip) > sqn (rt \<xi>) ip)))"
+  by inv_cterms (clarsimp split: if_split_asm option.split_asm)+
+
+text \<open>Proposition 7.27\<close>
+
+lemma route_tables_fresher:
+  "paodv i \<TTurnstile>\<^sub>A (recvmsg rreq_rrep_sn \<rightarrow>) onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)).
+                                                                \<forall>dip\<in>kD(rt \<xi>). rt \<xi> \<sqsubseteq>\<^bsub>dip\<^esub> rt \<xi>')"
+  proof (inv_cterms inv add:
+           onl_invariant_sterms [OF aodv_wf dests_vD_inc_sqn [THEN invariant_restrict_inD]]
+           onl_invariant_sterms [OF aodv_wf hop_count_positive [THEN invariant_restrict_inD]]
+           onl_invariant_sterms [OF aodv_wf osn_rreq]
+           onl_invariant_sterms [OF aodv_wf dsn_rrep]
+           onl_invariant_sterms [OF aodv_wf addpreRT_welldefined [THEN invariant_restrict_inD]])
+    fix \<xi> pp p'
+    assume "(\<xi>, pp) \<in> reachable (paodv i) (recvmsg rreq_rrep_sn)"
+       and "{PRreq-:2}\<lbrakk>\<lambda>\<xi>. \<xi>\<lparr>rt := update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})\<rparr>\<rbrakk>
+               p' \<in> sterms \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pp"
+       and "Suc 0 \<le> osn \<xi>"
+       and *: "\<forall>ip\<in>kD (rt \<xi>). Suc 0 \<le> the (dhops (rt \<xi>) ip)"
+    show "\<forall>ip\<in>kD (rt \<xi>). rt \<xi> \<sqsubseteq>\<^bsub>ip\<^esub> update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})"
+    proof
+      fix ip
+      assume "ip\<in>kD (rt \<xi>)"
+      moreover with * have "1 \<le> the (dhops (rt \<xi>) ip)" by simp
+      moreover from \<open>Suc 0 \<le> osn \<xi>\<close>
+        have "update_arg_wf (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})" ..
+      ultimately show "rt \<xi> \<sqsubseteq>\<^bsub>ip\<^esub> update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})"
+        by (rule rt_fresher_update)
+    qed
+  next
+    fix \<xi> pp p'
+    assume "(\<xi>, pp) \<in> reachable (paodv i) (recvmsg rreq_rrep_sn)"
+       and "{PRrep-:1}\<lbrakk>\<lambda>\<xi>. \<xi>\<lparr>rt := update (rt \<xi>) (dip \<xi>) (dsn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})\<rparr>\<rbrakk>
+            p' \<in> sterms \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pp"
+       and "Suc 0 \<le> dsn \<xi>"
+       and *: "\<forall>ip\<in>kD (rt \<xi>). Suc 0 \<le> the (dhops (rt \<xi>) ip)"
+    show "\<forall>ip\<in>kD (rt \<xi>). rt \<xi> \<sqsubseteq>\<^bsub>ip\<^esub> update (rt \<xi>) (dip \<xi>) (dsn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})"
+    proof
+      fix ip
+      assume "ip\<in>kD (rt \<xi>)"
+      moreover with * have "1 \<le> the (dhops (rt \<xi>) ip)" by simp
+      moreover from \<open>Suc 0 \<le> dsn \<xi>\<close>
+        have "update_arg_wf (dsn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})" ..
+      ultimately show "rt \<xi> \<sqsubseteq>\<^bsub>ip\<^esub> update (rt \<xi>) (dip \<xi>) (dsn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})"
+        by (rule rt_fresher_update)
+    qed
+  qed
+
+end
+

formal/afp/AODV/document/root.tex

@@ -0,0 +1,70 @@
+% vim:nojs:spelllang=en_au tw=76 sw=4 sts=4 fo+=awn fmr={-{,}-} et ts=8
+\documentclass[11pt,a4paper]{report}
+\usepackage[T1]{fontenc}
+\usepackage{isabelle,isabellesym}
+
+\usepackage{amssymb}
+\usepackage[only,bigsqcap]{stmaryrd}
+
+\usepackage{mathpartir}
+\usepackage[margin=10mm,bottom=15mm]{geometry}
+\usepackage[final]{graphicx}
+
+% this should be the last package used
+\usepackage{pdfsetup}
+
+% urls in roman style, theory text in math-similar italics
+\urlstyle{rm}
+\isabellestyle{rm}
+
+\begin{document}
+
+\title{Loop freedom of the (untimed) AODV routing protocol}
+\author{Timothy Bourke\thanks{Inria,
+                              \'Ecole normale sup\'erieure,
+                              and NICTA}
+        \and
+        Peter H\"ofner\thanks{NICTA
+                              and Computer Science and Engineering, UNSW}}
+\maketitle
+
+\begin{abstract}
+The Ad hoc On-demand Distance Vector (AODV) routing protocol~\cite{RFC3561} 
+allows the nodes in a Mobile Ad hoc Network (MANET) or a Wireless Mesh 
+Network (WMN) to know where to forward data packets. Such a protocol is 
+`loop free' if it never leads to routing decisions that forward packets in 
+circles.
+
+This development mechanises an existing pen-and-paper proof of loop freedom 
+of AODV~\cite{FehnkerEtAl:AWN:2013}.
+The protocol is modelled in the Algebra of Wireless Networks (AWN),
+which is the subject of an earlier paper~\cite{BourkeEtAl:MechAWN:2014} and 
+mechanization~\cite{Bourke14}.
+The proof relies on a novel compositional approach for lifting invariants to 
+networks of nodes.
+
+We exploit the mechanization to analyse several variants of AODV and show 
+that Isabelle/HOL can re-establish most proof obligations automatically and 
+identify exactly the steps that are no longer valid.
+Each of the variants is essentially a modified copy of the main development.
+
+Further documentation is available in~\cite{BourkevGlHof:ATVA:2014}.
+
+\centering{\includegraphics[width=\textwidth]{session_graph}}
+\end{abstract}
+
+\newpage
+\tableofcontents
+
+% sane default for proof documents
+\parindent 0pt\parskip 0.5ex
+
+% generated text of all theories
+\newpage
+\input{session}
+
+% optional bibliography
+\bibliographystyle{abbrv}
+\bibliography{root}
+
+\end{document}

formal/afp/AODV/variants/a_norreqid/A_Aodv.thy

@@ -0,0 +1,532 @@
+(*  Title:       variants/a_norreqid/Aodv.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+    Author:      Peter Höfner, NICTA
+*)
+
+section "The AODV protocol"
+
+theory A_Aodv
+imports A_Aodv_Data A_Aodv_Message
+        AWN.AWN_SOS_Labels AWN.AWN_Invariants
+begin
+
+subsection "Data state"
+
+record state =
+  ip    :: "ip"
+  sn    :: "sqn"
+  rt    :: "rt" 
+  rreqs :: "(ip \<times> sqn) set"
+  store :: "store"
+  (* all locals *)
+  msg    :: "msg"
+  data   :: "data"
+  dests  :: "ip \<rightharpoonup> sqn"
+  pre    :: "ip set"
+  dip    :: "ip"
+  oip    :: "ip"
+  hops   :: "nat"
+  dsn    :: "sqn"
+  dsk    :: "k"
+  osn    :: "sqn"
+  sip    :: "ip"
+
+abbreviation aodv_init :: "ip \<Rightarrow> state"
+where "aodv_init i \<equiv> \<lparr>
+         ip = i,
+         sn = 1,
+         rt = Map.empty,
+         rreqs = {},
+         store = Map.empty,
+
+         msg    = (SOME x. True),
+         data   = (SOME x. True),
+         dests  = (SOME x. True),
+         pre    = (SOME x. True),
+         dip    = (SOME x. True),
+         oip    = (SOME x. True),
+         hops   = (SOME x. True),
+         dsn    = (SOME x. True),
+         dsk    = (SOME x. True),
+         osn    = (SOME x. True),
+         sip    = (SOME x. x \<noteq> i)
+       \<rparr>"
+
+lemma some_neq_not_eq [simp]: "\<not>((SOME x :: nat. x \<noteq> i) = i)"
+  by (subst some_eq_ex) (metis zero_neq_numeral)
+
+definition clear_locals :: "state \<Rightarrow> state"
+where "clear_locals \<xi> = \<xi> \<lparr>
+    msg    := (SOME x. True),
+    data   := (SOME x. True),
+    dests  := (SOME x. True),
+    pre    := (SOME x. True),
+    dip    := (SOME x. True),
+    oip    := (SOME x. True),
+    hops   := (SOME x. True),
+    dsn    := (SOME x. True),
+    dsk    := (SOME x. True),
+    osn    := (SOME x. True),
+    sip    := (SOME x. x \<noteq> ip \<xi>)
+  \<rparr>"
+
+lemma clear_locals_sip_not_ip [simp]: "\<not>(sip (clear_locals \<xi>) = ip \<xi>)"
+  unfolding clear_locals_def by simp
+
+lemma clear_locals_but_not_globals [simp]:
+  "ip (clear_locals \<xi>) = ip \<xi>"
+  "sn (clear_locals \<xi>) = sn \<xi>"
+  "rt (clear_locals \<xi>) = rt \<xi>"
+  "rreqs (clear_locals \<xi>) = rreqs \<xi>"
+  "store (clear_locals \<xi>) = store \<xi>"
+  unfolding clear_locals_def by auto
+
+subsection "Auxilliary message handling definitions"
+
+definition is_newpkt
+where "is_newpkt \<xi> \<equiv> case msg \<xi> of
+                       Newpkt data' dip' \<Rightarrow> { \<xi>\<lparr>data := data', dip := dip'\<rparr> }
+                     | _ \<Rightarrow> {}"
+
+definition is_pkt
+where "is_pkt \<xi> \<equiv> case msg \<xi> of
+                    Pkt data' dip' oip' \<Rightarrow> { \<xi>\<lparr> data := data', dip := dip', oip := oip' \<rparr> }
+                  | _ \<Rightarrow> {}"
+
+definition is_rreq
+where "is_rreq \<xi> \<equiv> case msg \<xi> of
+                     Rreq hops' dip' dsn' dsk' oip' osn' sip' \<Rightarrow>
+                       { \<xi>\<lparr> hops := hops', dip := dip', dsn := dsn',
+                            dsk := dsk', oip := oip', osn := osn', sip := sip' \<rparr> }
+                   | _ \<Rightarrow> {}"
+
+lemma is_rreq_asm [dest!]:
+  assumes "\<xi>' \<in> is_rreq \<xi>"
+    shows "(\<exists>hops' rreqid' dip' dsn' dsk' oip' osn' sip'.
+               msg \<xi> = Rreq hops' dip' dsn' dsk' oip' osn' sip' \<and>
+               \<xi>' = \<xi>\<lparr> hops := hops', dip := dip', dsn := dsn',
+                       dsk := dsk', oip := oip', osn := osn', sip := sip' \<rparr>)"
+  using assms unfolding is_rreq_def
+  by (cases "msg \<xi>") simp_all
+
+definition is_rrep
+where "is_rrep \<xi> \<equiv> case msg \<xi> of
+                     Rrep hops' dip' dsn' oip' sip' \<Rightarrow>
+                       { \<xi>\<lparr> hops := hops', dip := dip', dsn := dsn', oip := oip', sip := sip' \<rparr> }
+                   | _ \<Rightarrow> {}"
+
+lemma is_rrep_asm [dest!]:
+  assumes "\<xi>' \<in> is_rrep \<xi>"
+    shows "(\<exists>hops' dip' dsn' oip' sip'.
+               msg \<xi> = Rrep hops' dip' dsn' oip' sip' \<and>
+               \<xi>' = \<xi>\<lparr> hops := hops', dip := dip', dsn := dsn', oip := oip', sip := sip' \<rparr>)"
+  using assms unfolding is_rrep_def
+  by (cases "msg \<xi>") simp_all
+
+definition is_rerr
+where "is_rerr \<xi> \<equiv> case msg \<xi> of
+                     Rerr dests' sip' \<Rightarrow> { \<xi>\<lparr> dests := dests', sip := sip' \<rparr> }
+                   | _ \<Rightarrow> {}"
+
+lemma is_rerr_asm [dest!]:
+  assumes "\<xi>' \<in> is_rerr \<xi>"
+    shows "(\<exists>dests' sip'.
+               msg \<xi> = Rerr dests' sip' \<and>
+               \<xi>' = \<xi>\<lparr> dests := dests', sip := sip' \<rparr>)"
+  using assms unfolding is_rerr_def
+  by (cases "msg \<xi>") simp_all
+
+lemmas is_msg_defs =
+  is_rerr_def is_rrep_def is_rreq_def is_pkt_def is_newpkt_def
+
+lemma is_msg_inv_ip [simp]:
+  "\<xi>' \<in> is_rerr \<xi>   \<Longrightarrow> ip \<xi>' = ip \<xi>"
+  "\<xi>' \<in> is_rrep \<xi>   \<Longrightarrow> ip \<xi>' = ip \<xi>"
+  "\<xi>' \<in> is_rreq \<xi>   \<Longrightarrow> ip \<xi>' = ip \<xi>"
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> ip \<xi>' = ip \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> ip \<xi>' = ip \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+lemma is_msg_inv_sn [simp]:
+  "\<xi>' \<in> is_rerr \<xi>   \<Longrightarrow> sn \<xi>' = sn \<xi>"
+  "\<xi>' \<in> is_rrep \<xi>   \<Longrightarrow> sn \<xi>' = sn \<xi>"
+  "\<xi>' \<in> is_rreq \<xi>   \<Longrightarrow> sn \<xi>' = sn \<xi>"
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> sn \<xi>' = sn \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> sn \<xi>' = sn \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+lemma is_msg_inv_rt [simp]:
+  "\<xi>' \<in> is_rerr \<xi>   \<Longrightarrow> rt \<xi>' = rt \<xi>"
+  "\<xi>' \<in> is_rrep \<xi>   \<Longrightarrow> rt \<xi>' = rt \<xi>"
+  "\<xi>' \<in> is_rreq \<xi>   \<Longrightarrow> rt \<xi>' = rt \<xi>"
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> rt \<xi>' = rt \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> rt \<xi>' = rt \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+lemma is_msg_inv_rreqs [simp]:
+  "\<xi>' \<in> is_rerr \<xi>   \<Longrightarrow> rreqs \<xi>' = rreqs \<xi>"
+  "\<xi>' \<in> is_rrep \<xi>   \<Longrightarrow> rreqs \<xi>' = rreqs \<xi>"
+  "\<xi>' \<in> is_rreq \<xi>   \<Longrightarrow> rreqs \<xi>' = rreqs \<xi>"
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> rreqs \<xi>' = rreqs \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> rreqs \<xi>' = rreqs \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+lemma is_msg_inv_store [simp]:
+  "\<xi>' \<in> is_rerr \<xi>   \<Longrightarrow> store \<xi>' = store \<xi>"
+  "\<xi>' \<in> is_rrep \<xi>   \<Longrightarrow> store \<xi>' = store \<xi>"
+  "\<xi>' \<in> is_rreq \<xi>   \<Longrightarrow> store \<xi>' = store \<xi>"
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> store \<xi>' = store \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> store \<xi>' = store \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+lemma is_msg_inv_sip [simp]:
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> sip \<xi>' = sip \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> sip \<xi>' = sip \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+subsection "The protocol process"
+
+datatype pseqp =
+    PAodv
+  | PNewPkt
+  | PPkt
+  | PRreq
+  | PRrep
+  | PRerr
+
+fun nat_of_seqp :: "pseqp \<Rightarrow> nat"
+where
+  "nat_of_seqp PAodv  = 1"
+| "nat_of_seqp PPkt   = 2"
+| "nat_of_seqp PNewPkt   = 3"
+| "nat_of_seqp PRreq  = 4"
+| "nat_of_seqp PRrep  = 5"
+| "nat_of_seqp PRerr  = 6"
+
+instantiation "pseqp" :: ord
+begin
+definition less_eq_seqp [iff]: "l1 \<le> l2 = (nat_of_seqp l1 \<le> nat_of_seqp l2)"
+definition less_seqp [iff]:    "l1 < l2 = (nat_of_seqp l1 < nat_of_seqp l2)"
+instance ..
+end
+
+abbreviation AODV
+where
+  "AODV \<equiv> \<lambda>_. \<lbrakk>clear_locals\<rbrakk> call(PAodv)"
+
+abbreviation PKT
+where
+  "PKT args \<equiv>
+
+     \<lbrakk>\<xi>. let (data, dip, oip) = args \<xi> in
+         (clear_locals \<xi>) \<lparr> data := data, dip := dip, oip := oip \<rparr>\<rbrakk>
+     call(PPkt)"
+abbreviation NEWPKT
+where
+  "NEWPKT args \<equiv>
+     \<lbrakk>\<xi>. let (data, dip) = args \<xi> in
+         (clear_locals \<xi>) \<lparr> data := data, dip := dip \<rparr>\<rbrakk>
+     call(PNewPkt)"
+
+abbreviation RREQ
+where
+  "RREQ args \<equiv>
+     \<lbrakk>\<xi>. let (hops, dip, dsn, dsk, oip, osn, sip) = args \<xi> in
+         (clear_locals \<xi>) \<lparr> hops := hops,  dip := dip,
+                            dsn := dsn, dsk := dsk, oip := oip,
+                            osn := osn, sip := sip \<rparr>\<rbrakk>
+     call(PRreq)"
+
+abbreviation RREP
+where
+  "RREP args \<equiv>
+     \<lbrakk>\<xi>. let (hops, dip, dsn, oip, sip) = args \<xi> in
+         (clear_locals \<xi>) \<lparr> hops := hops, dip := dip, dsn := dsn,
+                            oip := oip, sip := sip \<rparr>\<rbrakk>
+     call(PRrep)"
+
+abbreviation RERR
+where
+  "RERR args \<equiv>
+     \<lbrakk>\<xi>. let (dests, sip) = args \<xi> in
+         (clear_locals \<xi>) \<lparr> dests := dests, sip := sip \<rparr>\<rbrakk>
+     call(PRerr)"
+
+fun \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V :: "(state, msg, pseqp, pseqp label) seqp_env"
+where
+  "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PAodv = labelled PAodv (
+     receive(\<lambda>msg' \<xi>. \<xi> \<lparr> msg := msg' \<rparr>).
+     (    \<langle>is_newpkt\<rangle> NEWPKT(\<lambda>\<xi>. (data \<xi>, ip \<xi>))
+       \<oplus> \<langle>is_pkt\<rangle> PKT(\<lambda>\<xi>. (data \<xi>, dip \<xi>, oip \<xi>))
+       \<oplus> \<langle>is_rreq\<rangle>
+            \<lbrakk>\<xi>. \<xi> \<lparr>rt := update (rt \<xi>) (sip \<xi>) (0, unk, val, 1, sip \<xi>, {}) \<rparr>\<rbrakk>
+            RREQ(\<lambda>\<xi>. (hops \<xi>, dip \<xi>, dsn \<xi>, dsk \<xi>, oip \<xi>, osn \<xi>, sip \<xi>))
+       \<oplus> \<langle>is_rrep\<rangle>
+            \<lbrakk>\<xi>. \<xi> \<lparr>rt := update (rt \<xi>) (sip \<xi>) (0, unk, val, 1, sip \<xi>, {}) \<rparr>\<rbrakk>
+            RREP(\<lambda>\<xi>. (hops \<xi>, dip \<xi>, dsn \<xi>, oip \<xi>, sip \<xi>))
+       \<oplus> \<langle>is_rerr\<rangle>
+            \<lbrakk>\<xi>. \<xi> \<lparr>rt := update (rt \<xi>) (sip \<xi>) (0, unk, val, 1, sip \<xi>, {}) \<rparr>\<rbrakk>
+            RERR(\<lambda>\<xi>. (dests \<xi>, sip \<xi>))
+     )
+     \<oplus> \<langle>\<lambda>\<xi>. { \<xi>\<lparr> dip := dip \<rparr> | dip. dip \<in> qD(store \<xi>) \<inter> vD(rt \<xi>) }\<rangle>
+          \<lbrakk>\<xi>. \<xi> \<lparr> data := hd(\<sigma>\<^bsub>queue\<^esub>(store \<xi>, dip \<xi>)) \<rparr>\<rbrakk>
+          unicast(\<lambda>\<xi>. the (nhop (rt \<xi>) (dip \<xi>)), \<lambda>\<xi>. pkt(data \<xi>, dip \<xi>, ip \<xi>)).
+            \<lbrakk>\<xi>. \<xi> \<lparr> store := the (drop (dip \<xi>) (store \<xi>)) \<rparr>\<rbrakk>
+            AODV()
+          \<triangleright> \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if (rip \<in> vD (rt \<xi>) \<and> nhop (rt \<xi>) rip = nhop (rt \<xi>) (dip \<xi>))
+                                     then Some (inc (sqn (rt \<xi>) rip)) else None) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                                     then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+             groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)). AODV()
+     \<oplus> \<langle>\<lambda>\<xi>. { \<xi>\<lparr> dip := dip \<rparr>
+             | dip. dip \<in> qD(store \<xi>) - vD(rt \<xi>) \<and> the (\<sigma>\<^bsub>p-flag\<^esub>(store \<xi>, dip)) = req }\<rangle>
+         \<lbrakk>\<xi>. \<xi> \<lparr> store := unsetRRF (store \<xi>) (dip \<xi>) \<rparr>\<rbrakk>
+         \<lbrakk>\<xi>. \<xi> \<lparr> sn := inc (sn \<xi>) \<rparr>\<rbrakk>
+         \<lbrakk>\<xi>. \<xi> \<lparr> rreqs := rreqs \<xi> \<union> {(ip \<xi>, sn \<xi>)} \<rparr>\<rbrakk>
+         broadcast(\<lambda>\<xi>. rreq(0, dip \<xi>, sqn (rt \<xi>) (dip \<xi>), sqnf (rt \<xi>) (dip \<xi>),
+                            ip \<xi>, sn \<xi>, ip \<xi>)). AODV())"
+
+|  "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PNewPkt = labelled PNewPkt (
+     \<langle>\<xi>. dip \<xi> = ip \<xi>\<rangle>
+        deliver(\<lambda>\<xi>. data \<xi>).AODV()
+     \<oplus> \<langle>\<xi>. dip \<xi> \<noteq> ip \<xi>\<rangle>
+        \<lbrakk>\<xi>. \<xi> \<lparr> store := add (data \<xi>) (dip \<xi>) (store \<xi>) \<rparr>\<rbrakk>
+        AODV())"
+
+| "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PPkt = labelled PPkt (
+     \<langle>\<xi>. dip \<xi> = ip \<xi>\<rangle>
+        deliver(\<lambda>\<xi>. data \<xi>).AODV()
+     \<oplus> \<langle>\<xi>. dip \<xi> \<noteq> ip \<xi>\<rangle>
+     (
+       \<langle>\<xi>. dip \<xi> \<in> vD (rt \<xi>)\<rangle>
+         unicast(\<lambda>\<xi>. the (nhop (rt \<xi>) (dip \<xi>)), \<lambda>\<xi>. pkt(data \<xi>, dip \<xi>, oip \<xi>)).AODV()
+         \<triangleright>
+           \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if (rip \<in> vD (rt \<xi>) \<and> nhop (rt \<xi>) rip = nhop (rt \<xi>) (dip \<xi>))
+                                   then Some (inc (sqn (rt \<xi>) rip)) else None) \<rparr>\<rbrakk>
+           \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+           \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+           \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+           \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                                   then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+           groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)).AODV()
+       \<oplus> \<langle>\<xi>. dip \<xi> \<notin> vD (rt \<xi>)\<rangle>
+       (
+           \<langle>\<xi>. dip \<xi> \<in> iD (rt \<xi>)\<rangle>
+             groupcast(\<lambda>\<xi>. the (precs (rt \<xi>) (dip \<xi>)),
+                       \<lambda>\<xi>. rerr([dip \<xi> \<mapsto> sqn (rt \<xi>) (dip \<xi>)], ip \<xi>)). AODV()
+           \<oplus> \<langle>\<xi>. dip \<xi> \<notin> iD (rt \<xi>)\<rangle>
+              AODV()
+       )
+     ))"
+
+| "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRreq = labelled PRreq (
+     \<langle>\<xi>. (oip \<xi>, osn \<xi>) \<in> rreqs \<xi>\<rangle>
+       AODV()
+     \<oplus> \<langle>\<xi>. (oip \<xi>, osn \<xi>) \<notin> rreqs \<xi>\<rangle>
+       \<lbrakk>\<xi>. \<xi> \<lparr> rt := update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, hops \<xi> + 1, sip \<xi>, {}) \<rparr>\<rbrakk>
+       \<lbrakk>\<xi>. \<xi> \<lparr> rreqs := rreqs \<xi> \<union> {(oip \<xi>, osn \<xi>)} \<rparr>\<rbrakk>
+       (
+         \<langle>\<xi>. dip \<xi> = ip \<xi>\<rangle>
+           \<lbrakk>\<xi>. \<xi> \<lparr> sn := max (sn \<xi>) (dsn \<xi>) \<rparr>\<rbrakk>
+           unicast(\<lambda>\<xi>. the (nhop (rt \<xi>) (oip \<xi>)), \<lambda>\<xi>. rrep(0, dip \<xi>, sn \<xi>, oip \<xi>, ip \<xi>)).AODV()
+           \<triangleright>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if (rip \<in> vD (rt \<xi>) \<and> nhop (rt \<xi>) rip = nhop (rt \<xi>) (oip \<xi>))
+                                     then Some (inc (sqn (rt \<xi>) rip)) else None) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                                     then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+             groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)).AODV()
+         \<oplus> \<langle>\<xi>. dip \<xi> \<noteq> ip \<xi>\<rangle>
+         (
+           \<langle>\<xi>. dip \<xi> \<in> vD (rt \<xi>) \<and> dsn \<xi> \<le> sqn (rt \<xi>) (dip \<xi>) \<and> sqnf (rt \<xi>) (dip \<xi>) = kno\<rangle>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := the (addpreRT (rt \<xi>) (dip \<xi>) {sip \<xi>}) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := the (addpreRT (rt \<xi>) (oip \<xi>) {the (nhop (rt \<xi>) (dip \<xi>))}) \<rparr>\<rbrakk> 
+             unicast(\<lambda>\<xi>. the (nhop (rt \<xi>) (oip \<xi>)), \<lambda>\<xi>. rrep(the (dhops (rt \<xi>) (dip \<xi>)), dip \<xi>,
+                         sqn (rt \<xi>) (dip \<xi>), oip \<xi>, ip \<xi>)).
+             AODV()
+           \<triangleright>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if (rip \<in> vD (rt \<xi>) \<and> nhop (rt \<xi>) rip = nhop (rt \<xi>) (oip \<xi>))
+                                     then Some (inc (sqn (rt \<xi>) rip)) else None) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                                     then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+             groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)).AODV()
+           \<oplus> \<langle>\<xi>. dip \<xi> \<notin> vD (rt \<xi>) \<or> sqn (rt \<xi>) (dip \<xi>) < dsn \<xi> \<or> sqnf (rt \<xi>) (dip \<xi>) = unk\<rangle>
+             broadcast(\<lambda>\<xi>. rreq(hops \<xi> + 1, dip \<xi>, max (sqn (rt \<xi>) (dip \<xi>)) (dsn \<xi>),
+                                dsk \<xi>, oip \<xi>, osn \<xi>, ip \<xi>)).
+             AODV()
+         )
+       ))"
+
+| "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRrep = labelled PRrep (
+     \<langle>\<xi>. rt \<xi> \<noteq> update (rt \<xi>) (dip \<xi>) (dsn \<xi>, kno, val, hops \<xi> + 1, sip \<xi>, {}) \<rangle>
+     (
+       \<lbrakk>\<xi>. \<xi> \<lparr> rt := update (rt \<xi>) (dip \<xi>) (dsn \<xi>, kno, val, hops \<xi> + 1, sip \<xi>, {}) \<rparr> \<rbrakk>
+       (
+         \<langle>\<xi>. oip \<xi> = ip \<xi> \<rangle>
+            AODV()
+         \<oplus> \<langle>\<xi>. oip \<xi> \<noteq> ip \<xi> \<rangle>
+         (
+           \<langle>\<xi>. oip \<xi> \<in> vD (rt \<xi>)\<rangle>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := the (addpreRT (rt \<xi>) (dip \<xi>) {the (nhop (rt \<xi>) (oip \<xi>))}) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := the (addpreRT (rt \<xi>) (the (nhop (rt \<xi>) (dip \<xi>)))
+                                               {the (nhop (rt \<xi>) (oip \<xi>))}) \<rparr>\<rbrakk> 
+             unicast(\<lambda>\<xi>. the (nhop (rt \<xi>) (oip \<xi>)), \<lambda>\<xi>. rrep(hops \<xi> + 1, dip \<xi>, dsn \<xi>, oip \<xi>, ip \<xi>)).
+             AODV()
+           \<triangleright>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if (rip \<in> vD (rt \<xi>) \<and> nhop (rt \<xi>) rip = nhop (rt \<xi>) (oip \<xi>))
+                                     then Some (inc (sqn (rt \<xi>) rip)) else None) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                                     then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+             groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)).AODV()
+           \<oplus> \<langle>\<xi>. oip \<xi> \<notin> vD (rt \<xi>)\<rangle>
+             AODV()
+         )
+       )
+     )
+     \<oplus> \<langle>\<xi>. rt \<xi> = update (rt \<xi>) (dip \<xi>) (dsn \<xi>, kno, val, hops \<xi> + 1, sip \<xi>, {}) \<rangle>
+         AODV()
+     )"
+
+| "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRerr = labelled PRerr (
+     \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. case (dests \<xi>) rip of None \<Rightarrow> None
+                       | Some rsn \<Rightarrow> if rip \<in> vD (rt \<xi>) \<and> the (nhop (rt \<xi>) rip) = sip \<xi>
+                                       \<and> sqn (rt \<xi>) rip < rsn then Some rsn else None) \<rparr>\<rbrakk>
+     \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+     \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+     \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+     \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                             then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+     groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)). AODV())"
+
+declare \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V.simps [simp del, code del]
+lemmas \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_simps [simp, code] = \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V.simps [simplified]
+
+fun \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton
+where
+    "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PAodv   = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PAodv)"
+  | "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PNewPkt = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PNewPkt)"
+  | "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PPkt    = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PPkt)"
+  | "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PRreq   = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRreq)"
+  | "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PRrep   = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRrep)"
+  | "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PRerr   = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRerr)"
+
+lemma \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton_wf [simp]:
+  "wellformed \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton"
+  proof (rule, intro allI)
+    fix pn pn'
+    show "call(pn') \<notin> stermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton pn)"
+      by (cases pn) simp_all
+  qed
+
+declare \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton.simps [simp del, code del]
+lemmas \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton_simps [simp, code]
+           = \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton.simps [simplified \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_simps seqp_skeleton.simps]
+
+lemma aodv_proc_cases [dest]:
+  fixes p pn
+  shows "p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pn) \<Longrightarrow>
+                                (p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PAodv) \<or> 
+                                 p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PNewPkt)  \<or>
+                                 p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PPkt)  \<or>
+                                 p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRreq) \<or>
+                                 p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRrep) \<or>
+                                 p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRerr))"
+  by (cases pn) simp_all
+
+definition \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V :: "ip \<Rightarrow> (state \<times> (state, msg, pseqp, pseqp label) seqp) set"
+where "\<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i \<equiv> {(aodv_init i, \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PAodv)}"
+
+abbreviation paodv
+  :: "ip \<Rightarrow> (state \<times> (state, msg, pseqp, pseqp label) seqp, msg seq_action) automaton"
+where
+  "paodv i \<equiv> \<lparr> init = \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i, trans = seqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V \<rparr>"
+
+lemma aodv_trans: "trans (paodv i) = seqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V"
+  by simp
+
+lemma aodv_control_within [simp]: "control_within \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (init (paodv i))"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by (rule control_withinI) (auto simp del: \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_simps)
+
+lemma aodv_wf [simp]:
+  "wellformed \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V"
+  proof (rule, intro allI)
+    fix pn pn'
+    show "call(pn') \<notin> stermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pn)"
+      by (cases pn) simp_all
+  qed
+
+lemmas aodv_labels_not_empty [simp] = labels_not_empty [OF aodv_wf]
+
+lemma aodv_ex_label [intro]: "\<exists>l. l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p"
+  by (metis aodv_labels_not_empty all_not_in_conv)
+
+lemma aodv_ex_labelE [elim]:
+  assumes "\<forall>l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p. P l p"
+      and "\<exists>p l. P l p \<Longrightarrow> Q"
+    shows "Q"
+  using assms by (metis aodv_ex_label)
+
+lemma aodv_simple_labels [simp]: "simple_labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V"
+  proof
+    fix pn p
+    assume "p\<in>subterms(\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pn)"
+    thus "\<exists>!l. labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p = {l}"
+    by (cases pn) (simp_all cong: seqp_congs | elim disjE)+
+  qed
+
+lemma \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_labels [simp]: "(\<xi>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i \<Longrightarrow>  labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p = {PAodv-:0}"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by simp
+
+lemma aodv_init_kD_empty [simp]:
+  "(\<xi>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i \<Longrightarrow> kD (rt \<xi>) = {}"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def kD_def by simp
+
+lemma aodv_init_sip_not_ip [simp]: "\<not>(sip (aodv_init i) = i)" by simp
+
+lemma aodv_init_sip_not_ip' [simp]:
+  assumes "(\<xi>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i"
+    shows "sip \<xi> \<noteq> ip \<xi>"
+  using assms unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by simp
+
+lemma aodv_init_sip_not_i [simp]:
+  assumes "(\<xi>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i"
+    shows "sip \<xi> \<noteq> i"
+  using assms unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by simp
+
+lemma clear_locals_sip_not_ip':
+  assumes "ip \<xi> = i"
+    shows "\<not>(sip (clear_locals \<xi>) = i)"
+  using assms by auto
+
+text \<open>Stop the simplifier from descending into process terms.\<close>
+declare seqp_congs [cong]
+
+text \<open>Configure the main invariant tactic for AODV.\<close>
+
+declare
+  \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_simps [cterms_env]
+  aodv_proc_cases [ctermsl_cases]
+  seq_invariant_ctermsI [OF aodv_wf aodv_control_within aodv_simple_labels aodv_trans,
+                            cterms_intros]
+  seq_step_invariant_ctermsI [OF aodv_wf aodv_control_within aodv_simple_labels aodv_trans,
+                                 cterms_intros]
+
+end

formal/afp/AODV/variants/a_norreqid/A_Aodv_Data.thy

@@ -0,0 +1,986 @@
+(*  Title:       variants/a_norreqid/Aodv_Data.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+    Author:      Peter Höfner, NICTA
+*)
+
+section "Predicates and functions used in the AODV model"
+
+theory A_Aodv_Data
+imports A_Norreqid
+begin
+
+subsection "Sequence Numbers"
+
+text \<open>Sequence numbers approximate the relative freshness of routing information.\<close>
+
+definition inc :: "sqn \<Rightarrow> sqn"
+  where "inc sn \<equiv> if sn = 0 then sn else sn + 1"
+
+lemma less_than_inc [simp]: "x \<le> inc x"
+  unfolding inc_def by simp
+
+lemma inc_minus_suc_0 [simp]:
+  "inc x - Suc 0 = x"
+  unfolding inc_def by simp
+
+lemma inc_never_one' [simp, intro]: "inc x \<noteq> Suc 0"
+  unfolding inc_def by simp
+
+lemma inc_never_one [simp, intro]: "inc x \<noteq> 1"
+  by simp
+
+subsection "Modelling Routes"
+
+text \<open>
+  A route is a 6-tuple, @{term "(dsn, dsk, flag, hops, nhip, pre)"} where
+  @{term dsn} is the `destination sequence number', @{term dsk} is the
+  `destination-sequence-number status', @{term flag} is the route status,
+  @{term hops} is the number of hops to the destination, @{term nhip} is the
+  next hop toward the destination, and @{term pre} is the set of `precursor nodes'--those
+  interested in hearing about changes to the route.
+\<close>
+
+type_synonym r = "sqn \<times> k \<times> f \<times> nat \<times> ip \<times> ip set"
+
+definition proj2 :: "r \<Rightarrow> sqn" ("\<pi>\<^sub>2")
+  where "\<pi>\<^sub>2 \<equiv> \<lambda>(dsn, _, _, _, _, _). dsn"
+
+definition proj3 :: "r \<Rightarrow> k" ("\<pi>\<^sub>3")
+  where "\<pi>\<^sub>3 \<equiv> \<lambda>(_, dsk, _, _, _, _). dsk"
+
+definition proj4 :: "r \<Rightarrow> f" ("\<pi>\<^sub>4")
+  where "\<pi>\<^sub>4 \<equiv> \<lambda>(_, _, flag, _, _, _). flag"
+
+definition proj5 :: "r \<Rightarrow> nat" ("\<pi>\<^sub>5")
+  where "\<pi>\<^sub>5 \<equiv> \<lambda>(_, _, _, hops, _, _). hops"
+
+definition proj6 :: "r \<Rightarrow> ip" ("\<pi>\<^sub>6")
+  where "\<pi>\<^sub>6 \<equiv> \<lambda>(_, _, _, _, nhip, _). nhip"
+
+definition proj7 :: "r \<Rightarrow> ip set" ("\<pi>\<^sub>7")
+  where "\<pi>\<^sub>7 \<equiv> \<lambda>(_, _, _, _, _, pre). pre"
+
+lemma projs [simp]:
+  "\<pi>\<^sub>2(dsn, dsk, flag, hops, nhip, pre) = dsn"
+  "\<pi>\<^sub>3(dsn, dsk, flag, hops, nhip, pre) = dsk"
+  "\<pi>\<^sub>4(dsn, dsk, flag, hops, nhip, pre) = flag"
+  "\<pi>\<^sub>5(dsn, dsk, flag, hops, nhip, pre) = hops"
+  "\<pi>\<^sub>6(dsn, dsk, flag, hops, nhip, pre) = nhip"
+  "\<pi>\<^sub>7(dsn, dsk, flag, hops, nhip, pre) = pre"
+  by (clarsimp simp: proj2_def proj3_def proj4_def
+                     proj5_def proj6_def proj7_def)+
+
+lemma proj3_pred [intro]: "\<lbrakk> P kno; P unk \<rbrakk> \<Longrightarrow> P (\<pi>\<^sub>3 x)"
+  by (rule k.induct)
+
+lemma proj4_pred [intro]: "\<lbrakk> P val; P inv \<rbrakk> \<Longrightarrow> P (\<pi>\<^sub>4 x)"
+  by (rule f.induct)
+
+lemma proj6_pair_snd [simp]:
+  fixes dsn' r
+  shows "\<pi>\<^sub>6 (dsn', snd (r)) = \<pi>\<^sub>6(r)"
+  by (cases r) simp
+
+subsection "Routing Tables"
+
+text \<open>Routing tables map ip addresses to route entries.\<close>
+
+type_synonym rt = "ip \<rightharpoonup> r"
+
+syntax
+  "_Sigma_route" :: "rt \<Rightarrow> ip \<rightharpoonup> r"  ("\<sigma>\<^bsub>route\<^esub>'(_, _')")
+
+translations
+ "\<sigma>\<^bsub>route\<^esub>(rt, dip)" => "rt dip" 
+
+definition sqn :: "rt \<Rightarrow> ip \<Rightarrow> sqn"
+  where "sqn rt dip \<equiv> case \<sigma>\<^bsub>route\<^esub>(rt, dip) of Some r \<Rightarrow> \<pi>\<^sub>2(r) | None \<Rightarrow> 0"
+
+definition sqnf :: "rt \<Rightarrow> ip \<Rightarrow> k"
+  where "sqnf rt dip \<equiv> case \<sigma>\<^bsub>route\<^esub>(rt, dip) of Some r \<Rightarrow> \<pi>\<^sub>3(r) | None \<Rightarrow> unk"
+
+abbreviation flag :: "rt \<Rightarrow> ip \<rightharpoonup> f"
+  where "flag rt dip \<equiv> map_option \<pi>\<^sub>4 (\<sigma>\<^bsub>route\<^esub>(rt, dip))"
+
+abbreviation dhops :: "rt \<Rightarrow> ip \<rightharpoonup> nat"
+   where "dhops rt dip \<equiv> map_option \<pi>\<^sub>5 (\<sigma>\<^bsub>route\<^esub>(rt, dip))"
+
+abbreviation nhop :: "rt \<Rightarrow> ip \<rightharpoonup> ip"
+   where "nhop rt dip \<equiv> map_option \<pi>\<^sub>6 (\<sigma>\<^bsub>route\<^esub>(rt, dip))"
+
+abbreviation precs :: "rt \<Rightarrow> ip \<rightharpoonup> ip set"
+   where "precs rt dip \<equiv> map_option \<pi>\<^sub>7 (\<sigma>\<^bsub>route\<^esub>(rt, dip))"
+
+definition vD :: "rt \<Rightarrow> ip set"
+  where "vD rt \<equiv> {dip. flag rt dip = Some val}"
+
+definition iD :: "rt \<Rightarrow> ip set"
+  where "iD rt \<equiv> {dip. flag rt dip = Some inv}"
+
+definition kD :: "rt \<Rightarrow> ip set"
+  where "kD rt \<equiv> {dip. rt dip \<noteq> None}"
+
+lemma kD_is_vD_and_iD: "kD rt = vD rt \<union> iD rt"
+  unfolding kD_def vD_def iD_def by auto
+
+lemma vD_iD_gives_kD [simp]:
+   "\<And>ip rt. ip \<in> vD rt \<Longrightarrow> ip \<in> kD rt"
+   "\<And>ip rt. ip \<in> iD rt \<Longrightarrow> ip \<in> kD rt"
+  unfolding kD_is_vD_and_iD by simp_all
+
+lemma kD_Some [dest]:
+    fixes dip rt
+  assumes "dip \<in> kD rt"
+    shows "\<exists>dsn dsk flag hops nhip pre.
+           \<sigma>\<^bsub>route\<^esub>(rt, dip) = Some (dsn, dsk, flag, hops, nhip, pre)"
+  using assms unfolding kD_def by simp
+
+lemma kD_None [dest]:
+    fixes dip rt
+  assumes "dip \<notin> kD rt"
+    shows "\<sigma>\<^bsub>route\<^esub>(rt, dip) = None"
+  using assms unfolding kD_def
+  by (metis (mono_tags) mem_Collect_eq)
+
+lemma vD_Some [dest]:
+    fixes dip rt
+  assumes "dip \<in> vD rt"
+    shows "\<exists>dsn dsk hops nhip pre.
+           \<sigma>\<^bsub>route\<^esub>(rt, dip) = Some (dsn, dsk, val, hops, nhip, pre)"
+  using assms unfolding vD_def by simp
+
+lemma vD_empty [simp]: "vD Map.empty = {}"
+  unfolding vD_def by simp
+
+lemma iD_Some [dest]:
+    fixes dip rt
+  assumes "dip \<in> iD rt"
+    shows "\<exists>dsn dsk hops nhip pre.
+           \<sigma>\<^bsub>route\<^esub>(rt, dip) = Some (dsn, dsk, inv, hops, nhip, pre)"
+  using assms unfolding iD_def by simp
+
+lemma val_is_vD [elim]:
+    fixes ip rt
+  assumes "ip\<in>kD(rt)"
+      and "the (flag rt ip) = val"
+    shows "ip\<in>vD(rt)"
+  using assms unfolding vD_def by auto
+
+lemma inv_is_iD [elim]:
+    fixes ip rt
+  assumes "ip\<in>kD(rt)"
+      and "the (flag rt ip) = inv"
+    shows "ip\<in>iD(rt)"
+  using assms unfolding iD_def by auto
+
+lemma iD_flag_is_inv [elim, simp]:
+    fixes ip rt
+  assumes "ip\<in>iD(rt)"
+    shows "the (flag rt ip) = inv"
+  proof -
+    from \<open>ip\<in>iD(rt)\<close> have "ip\<in>kD(rt)" by auto
+    with assms show ?thesis unfolding iD_def by auto
+  qed
+
+lemma kD_but_not_vD_is_iD [elim]:
+    fixes ip rt
+  assumes "ip\<in>kD(rt)"
+      and "ip\<notin>vD(rt)"
+    shows "ip\<in>iD(rt)"
+  proof -
+    from \<open>ip\<in>kD(rt)\<close> obtain dsn dsk f hops nhop pre
+      where rtip: "rt ip = Some (dsn, dsk, f, hops, nhop, pre)"
+       by (metis kD_Some)
+    from \<open>ip\<notin>vD(rt)\<close> have "f \<noteq> val"
+    proof (rule contrapos_nn)
+      assume "f = val"
+      with rtip have "the (flag rt ip) = val" by simp
+      with \<open>ip\<in>kD(rt)\<close> show "ip\<in>vD(rt)" ..
+    qed
+    with rtip have "the (flag rt ip)= inv" by simp  
+    with \<open>ip\<in>kD(rt)\<close> show "ip\<in>iD(rt)" ..
+  qed
+
+lemma vD_or_iD [elim]:
+    fixes ip rt
+  assumes "ip\<in>kD(rt)"
+      and "ip\<in>vD(rt) \<Longrightarrow> P rt ip"
+      and "ip\<in>iD(rt) \<Longrightarrow> P rt ip"
+    shows "P rt ip"
+  proof -
+    from \<open>ip\<in>kD(rt)\<close> have "ip\<in>vD(rt) \<union> iD(rt)"
+      by (simp add: kD_is_vD_and_iD)
+    thus ?thesis by (auto elim: assms(2-3))
+  qed
+
+lemma proj5_eq_dhops: "\<And>dip rt. dip\<in>kD(rt) \<Longrightarrow> \<pi>\<^sub>5(the (rt dip)) = the (dhops rt dip)"
+  unfolding sqn_def by (drule kD_Some) clarsimp
+
+lemma proj4_eq_flag: "\<And>dip rt. dip\<in>kD(rt) \<Longrightarrow> \<pi>\<^sub>4(the (rt dip)) = the (flag rt dip)"
+  unfolding sqn_def by (drule kD_Some) clarsimp
+
+lemma proj2_eq_sqn: "\<And>dip rt. dip\<in>kD(rt) \<Longrightarrow> \<pi>\<^sub>2(the (rt dip)) = sqn rt dip"
+  unfolding sqn_def by (drule kD_Some) clarsimp
+
+lemma kD_sqnf_is_proj3 [simp]:
+  "\<And>ip rt. ip\<in>kD(rt) \<Longrightarrow> sqnf rt ip = \<pi>\<^sub>3(the (rt ip))"
+  unfolding sqnf_def by auto
+
+lemma vD_flag_val [simp]:
+  "\<And>dip rt. dip \<in> vD (rt) \<Longrightarrow> the (flag rt dip) = val"
+  unfolding vD_def by clarsimp
+
+lemma kD_update [simp]:
+  "\<And>rt nip v. kD (rt(nip \<mapsto> v)) = insert nip (kD rt)"
+  unfolding kD_def by auto
+
+lemma kD_empty [simp]: "kD Map.empty = {}"
+  unfolding kD_def by simp
+
+lemma ip_equal_or_known [elim]:
+  fixes rt ip ip'
+  assumes "ip = ip' \<or> ip\<in>kD(rt)"
+      and "ip = ip' \<Longrightarrow> P rt ip ip'"
+      and "\<lbrakk> ip \<noteq> ip'; ip\<in>kD(rt)\<rbrakk> \<Longrightarrow> P rt ip ip'"
+    shows "P rt ip ip'"
+  using assms by auto
+
+subsection "Updating Routing Tables"
+
+text \<open>Routing table entries are modified through explicit functions.
+      The properties of these functions are important in invariant proofs.\<close>
+
+subsubsection "Updating Precursor Lists"
+
+definition addpre :: "r \<Rightarrow> ip set \<Rightarrow> r"
+  where "addpre r npre \<equiv> let (dsn, dsk, flag, hops, nhip, pre) = r in
+                          (dsn, dsk, flag, hops, nhip, pre \<union> npre)"
+
+lemma proj2_addpre:
+  fixes v pre
+  shows "\<pi>\<^sub>2(addpre v pre) = \<pi>\<^sub>2(v)"
+  unfolding addpre_def by (cases v) simp
+
+lemma proj3_addpre:
+  fixes v pre
+  shows "\<pi>\<^sub>3(addpre v pre) = \<pi>\<^sub>3(v)"
+  unfolding addpre_def by (cases v) simp
+
+lemma proj4_addpre:
+  fixes v pre
+  shows "\<pi>\<^sub>4(addpre v pre) = \<pi>\<^sub>4(v)"
+  unfolding addpre_def by (cases v) simp
+
+lemma proj5_addpre:
+  fixes v pre
+  shows "\<pi>\<^sub>5(addpre v pre) = \<pi>\<^sub>5(v)"
+  unfolding addpre_def by (cases v) simp
+
+lemma proj6_addpre:
+  fixes dsn dsk flag hops nhip pre npre
+  shows "\<pi>\<^sub>6(addpre v npre) = \<pi>\<^sub>6(v)"
+  unfolding addpre_def by (cases v) simp
+
+lemma proj7_addpre:
+  fixes dsn dsk flag hops nhip pre npre
+  shows "\<pi>\<^sub>7(addpre v npre) = \<pi>\<^sub>7(v) \<union> npre"
+  unfolding addpre_def by (cases v) simp
+
+lemma addpre_empty: "addpre r {} = r"
+  unfolding addpre_def by simp
+
+lemma addpre_r:
+  "addpre (dsn, dsk, fl, hops, nhip, pre) npre = (dsn, dsk, fl, hops, nhip, pre \<union> npre)"
+  unfolding addpre_def by simp
+
+lemmas addpre_simps [simp] = proj2_addpre proj3_addpre proj4_addpre proj5_addpre
+                             proj6_addpre proj7_addpre addpre_empty addpre_r
+
+definition addpreRT :: "rt \<Rightarrow> ip \<Rightarrow> ip set \<rightharpoonup> rt"
+  where "addpreRT rt dip npre \<equiv>
+           map_option (\<lambda>s. rt (dip \<mapsto> addpre s npre)) (\<sigma>\<^bsub>route\<^esub>(rt, dip))"
+
+lemma snd_addpre [simp]:
+  "\<And>dsn dsn' v pre. (dsn, snd(addpre (dsn', v) pre)) = addpre (dsn, v) pre"
+  unfolding addpre_def by clarsimp
+
+lemma proj2_addpreRT [simp]:
+    fixes ip rt ip' npre
+  assumes "ip\<in>kD rt"
+      and "ip'\<in>kD rt"
+    shows "\<pi>\<^sub>2(the (the (addpreRT rt ip' npre) ip)) = \<pi>\<^sub>2(the (rt ip))"
+  using assms [THEN kD_Some] unfolding addpreRT_def by clarsimp
+
+lemma proj3_addpreRT [simp]:
+    fixes ip rt ip' npre
+  assumes "ip\<in>kD rt"
+      and "ip'\<in>kD rt"
+    shows "\<pi>\<^sub>3(the (the (addpreRT rt ip' npre) ip)) = \<pi>\<^sub>3(the (rt ip))"
+  using assms [THEN kD_Some] unfolding addpreRT_def by clarsimp
+
+lemma proj5_addpreRT [simp]:
+  "\<And>rt dip ip npre. dip\<in>kD(rt) \<Longrightarrow> \<pi>\<^sub>5(the (the (addpreRT rt dip npre) ip)) = \<pi>\<^sub>5(the (rt ip))"
+  unfolding addpreRT_def by auto
+
+lemma flag_addpreRT [simp]:
+    fixes rt pre ip dip
+  assumes "dip \<in> kD rt"
+    shows "flag (the (addpreRT rt dip pre)) ip = flag rt ip"
+  unfolding addpreRT_def
+  using assms [THEN kD_Some] by (clarsimp)
+
+  lemma kD_addpreRT [simp]:
+  fixes rt dip npre
+  assumes "dip \<in> kD rt"
+  shows "kD (the (addpreRT rt dip npre)) = kD rt"
+  unfolding kD_def addpreRT_def
+  using assms [THEN kD_Some]
+  by clarsimp blast
+
+lemma vD_addpreRT [simp]:
+  fixes rt dip npre
+  assumes "dip \<in> kD rt"
+  shows "vD (the (addpreRT rt dip npre)) = vD rt"
+  unfolding vD_def addpreRT_def
+  using assms [THEN kD_Some] by clarsimp auto
+
+lemma iD_addpreRT [simp]:
+  fixes rt dip npre
+  assumes "dip \<in> kD rt"
+  shows "iD (the (addpreRT rt dip npre)) = iD rt"
+  unfolding iD_def addpreRT_def
+  using assms [THEN kD_Some] by clarsimp auto
+
+lemma nhop_addpreRT [simp]:
+    fixes rt pre ip dip
+  assumes "dip \<in> kD rt"
+    shows "nhop (the (addpreRT rt dip pre)) ip = nhop rt ip"
+  unfolding sqn_def addpreRT_def
+  using assms [THEN kD_Some] by (clarsimp)
+
+lemma sqn_addpreRT [simp]:
+    fixes rt pre ip dip
+  assumes "dip \<in> kD rt"
+    shows "sqn (the (addpreRT rt dip pre)) ip = sqn rt ip"
+  unfolding sqn_def addpreRT_def
+  using assms [THEN kD_Some] by (clarsimp)
+
+lemma dhops_addpreRT [simp]:
+    fixes rt pre ip dip
+  assumes "dip \<in> kD rt"
+    shows "dhops (the (addpreRT rt dip pre)) ip = dhops rt ip"
+  unfolding addpreRT_def
+  using assms [THEN kD_Some] by (clarsimp)
+
+lemma sqnf_addpreRT [simp]:
+  "\<And>ip dip. ip\<in>kD(rt \<xi>) \<Longrightarrow> sqnf (the (addpreRT (rt \<xi>) ip npre)) dip = sqnf (rt \<xi>) dip"
+  unfolding sqnf_def addpreRT_def by auto
+
+subsubsection "Updating route entries"
+
+lemma in_kD_case [simp]:
+    fixes dip rt
+  assumes "dip \<in> kD(rt)"
+    shows "(case rt dip of None \<Rightarrow> en | Some r \<Rightarrow> es r) = es (the (rt dip))"
+  using assms [THEN kD_Some] by auto
+
+lemma not_in_kD_case [simp]:
+    fixes dip rt
+  assumes "dip \<notin> kD(rt)"
+    shows "(case rt dip of None \<Rightarrow> en | Some r \<Rightarrow> es r) = en"
+  using assms [THEN kD_None] by auto
+
+lemma rt_Some_sqn [dest]:
+    fixes rt and ip dsn dsk flag hops nhip pre
+  assumes "rt ip = Some (dsn, dsk, flag, hops, nhip, pre)"
+    shows "sqn rt ip = dsn"
+  unfolding sqn_def using assms by simp
+
+lemma not_kD_sqn [simp]:
+    fixes dip rt
+  assumes "dip \<notin> kD(rt)"
+    shows "sqn rt dip = 0"
+  using assms unfolding sqn_def
+  by simp
+
+definition update_arg_wf :: "r \<Rightarrow> bool"
+where "update_arg_wf r \<equiv> \<pi>\<^sub>4(r) = val \<and>
+                         (\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk) \<and>
+                         (\<pi>\<^sub>3(r) = unk \<longrightarrow> \<pi>\<^sub>5(r) = 1)"
+
+lemma update_arg_wf_gives_cases:
+  "\<And>r. update_arg_wf r \<Longrightarrow> (\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)"
+  unfolding update_arg_wf_def by simp
+
+lemma update_arg_wf_tuples [simp]:
+  "\<And>nhip pre. update_arg_wf (0, unk, val, Suc 0, nhip, pre)"
+  "\<And>n hops nhip pre. update_arg_wf (Suc n, kno, val, hops,  nhip, pre)"
+  unfolding update_arg_wf_def by auto
+
+lemma update_arg_wf_tuples' [elim]:
+  "\<And>n hops nhip pre. Suc 0 \<le> n \<Longrightarrow> update_arg_wf (n, kno, val, hops,  nhip, pre)"
+  unfolding update_arg_wf_def by auto
+
+lemma wf_r_cases [intro]:
+    fixes P r
+  assumes "update_arg_wf r"
+      and c1: "\<And>nhip pre. P (0, unk, val, Suc 0, nhip, pre)"
+      and c2: "\<And>dsn hops nhip pre. dsn > 0 \<Longrightarrow> P (dsn, kno, val, hops, nhip, pre)"
+    shows "P r"
+  proof -
+    obtain dsn dsk flag hops nhip pre
+    where *: "r = (dsn, dsk, flag, hops, nhip, pre)" by (cases r)
+    with \<open>update_arg_wf r\<close> have wf1: "flag = val"
+                            and wf2: "(dsn = 0) = (dsk = unk)"
+                            and wf3: "dsk = unk \<longrightarrow> (hops = 1)"
+      unfolding update_arg_wf_def by auto
+    have "P (dsn, dsk, flag, hops, nhip, pre)"
+    proof (cases dsk)
+      assume "dsk = unk"
+      moreover with wf2 wf3 have "dsn = 0" and "hops = Suc 0" by auto
+      ultimately show ?thesis using \<open>flag = val\<close> by simp (rule c1)
+    next
+      assume "dsk = kno"
+      moreover with wf2 have "dsn > 0" by simp
+      ultimately show ?thesis using \<open>flag = val\<close> by simp (rule c2)
+    qed
+    with * show "P r" by simp
+  qed
+
+definition update :: "rt \<Rightarrow> ip \<Rightarrow> r \<Rightarrow> rt"
+  where
+  "update rt ip r \<equiv>
+     case \<sigma>\<^bsub>route\<^esub>(rt, ip) of
+       None \<Rightarrow> rt (ip \<mapsto> r)
+     | Some s \<Rightarrow>
+          if \<pi>\<^sub>2(s) < \<pi>\<^sub>2(r) then rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(s)))
+          else if \<pi>\<^sub>2(s) = \<pi>\<^sub>2(r) \<and> (\<pi>\<^sub>5(s) > \<pi>\<^sub>5(r) \<or> \<pi>\<^sub>4(s) = inv)
+               then rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(s)))
+               else if \<pi>\<^sub>3(r) = unk
+                    then rt (ip \<mapsto> (\<pi>\<^sub>2(s), snd (addpre r (\<pi>\<^sub>7(s)))))
+                    else rt (ip \<mapsto> addpre s (\<pi>\<^sub>7(r)))"
+
+lemma update_simps [simp]:
+  fixes r s nrt nr nr' ns rt ip
+  defines "s \<equiv> the \<sigma>\<^bsub>route\<^esub>(rt, ip)"
+      and "nr \<equiv> addpre r (\<pi>\<^sub>7(s))"
+      and "nr' \<equiv> (\<pi>\<^sub>2(s), \<pi>\<^sub>3(nr), \<pi>\<^sub>4(nr), \<pi>\<^sub>5(nr), \<pi>\<^sub>6(nr), \<pi>\<^sub>7(nr))"
+      and "ns \<equiv> addpre s (\<pi>\<^sub>7(r))"
+  shows
+  "\<lbrakk>ip \<notin> kD(rt)\<rbrakk>                            \<Longrightarrow> update rt ip r = rt (ip \<mapsto> r)"
+  "\<lbrakk>ip \<in> kD(rt); sqn rt ip < \<pi>\<^sub>2(r)\<rbrakk>         \<Longrightarrow> update rt ip r = rt (ip \<mapsto> nr)"
+  "\<lbrakk>ip \<in> kD(rt); sqn rt ip = \<pi>\<^sub>2(r);
+                 the (dhops rt ip) > \<pi>\<^sub>5(r)\<rbrakk> \<Longrightarrow> update rt ip r = rt (ip \<mapsto> nr)"
+  "\<lbrakk>ip \<in> kD(rt); sqn rt ip = \<pi>\<^sub>2(r);
+                 flag rt ip = Some inv\<rbrakk>     \<Longrightarrow> update rt ip r = rt (ip \<mapsto> nr)"
+  "\<lbrakk>ip \<in> kD(rt); \<pi>\<^sub>3(r) = unk; (\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)\<rbrakk>  \<Longrightarrow> update rt ip r = rt (ip \<mapsto> nr')"
+  "\<lbrakk>ip \<in> kD(rt); sqn rt ip \<ge> \<pi>\<^sub>2(r); \<pi>\<^sub>3(r) = kno;
+    sqn rt ip = \<pi>\<^sub>2(r) \<Longrightarrow> the (dhops rt ip) \<le> \<pi>\<^sub>5(r) \<and> the (flag rt ip) = val \<rbrakk>
+                                            \<Longrightarrow> update rt ip r = rt (ip \<mapsto> ns)"
+  proof -
+    assume "ip\<notin>kD(rt)"
+    hence "\<sigma>\<^bsub>route\<^esub>(rt, ip) = None" ..
+    thus "update rt ip r = rt (ip \<mapsto> r)"
+      unfolding update_def by simp
+  next
+    assume "ip \<in> kD(rt)"
+       and "sqn rt ip < \<pi>\<^sub>2(r)"
+    from this(1) obtain dsn dsk fl hops nhip pre
+      where "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)
+    with \<open>sqn rt ip < \<pi>\<^sub>2(r)\<close> show "update rt ip r = rt (ip \<mapsto> nr)"
+      unfolding update_def nr_def s_def by auto
+  next
+    assume "ip \<in> kD(rt)"
+       and "sqn rt ip = \<pi>\<^sub>2(r)"
+       and "the (dhops rt ip) > \<pi>\<^sub>5(r)"
+    from this(1) obtain dsn dsk fl hops nhip pre
+      where "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)
+    with \<open>sqn rt ip = \<pi>\<^sub>2(r)\<close> and \<open>the (dhops rt ip) > \<pi>\<^sub>5(r)\<close>
+      show "update rt ip r = rt (ip \<mapsto> nr)"
+        unfolding update_def nr_def s_def by auto
+   next
+     assume "ip \<in> kD(rt)"
+        and "sqn rt ip = \<pi>\<^sub>2(r)"
+        and "flag rt ip = Some inv"
+    from this(1) obtain dsn dsk fl hops nhip pre
+      where "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)        
+     with \<open>sqn rt ip = \<pi>\<^sub>2(r)\<close> and \<open>flag rt ip = Some inv\<close>
+      show "update rt ip r = rt (ip \<mapsto> nr)"
+        unfolding update_def nr_def s_def by auto
+   next
+    assume "ip \<in> kD(rt)"
+       and "\<pi>\<^sub>3(r) = unk"
+       and "(\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)"
+    from this(1) obtain dsn dsk fl hops nhip pre
+      where "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)
+    with \<open>(\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)\<close> and \<open>\<pi>\<^sub>3(r) = unk\<close>
+      show "update rt ip r = rt (ip \<mapsto> nr')"
+        unfolding update_def nr'_def nr_def s_def
+      by (cases r) simp
+   next
+    assume "ip \<in> kD(rt)"
+       and otherassms: "sqn rt ip \<ge> \<pi>\<^sub>2(r)"
+           "\<pi>\<^sub>3(r) = kno"
+           "sqn rt ip = \<pi>\<^sub>2(r) \<Longrightarrow> the (dhops rt ip) \<le> \<pi>\<^sub>5(r) \<and> the (flag rt ip) = val"
+    from this(1) obtain dsn dsk fl hops nhip pre
+      where "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)
+     with otherassms show "update rt ip r = rt (ip \<mapsto> ns)"
+      unfolding update_def ns_def s_def by auto
+  qed
+
+lemma update_cases [elim]:
+  assumes "(\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)"
+      and c1: "\<lbrakk>ip \<notin> kD(rt)\<rbrakk> \<Longrightarrow> P (rt (ip \<mapsto> r))"
+
+      and c2: "\<lbrakk>ip \<in> kD(rt); sqn rt ip < \<pi>\<^sub>2(r)\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c3: "\<lbrakk>ip \<in> kD(rt); sqn rt ip = \<pi>\<^sub>2(r); the (dhops rt ip) > \<pi>\<^sub>5(r)\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c4: "\<lbrakk>ip \<in> kD(rt); sqn rt ip = \<pi>\<^sub>2(r); the (flag rt ip) = inv\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c5: "\<lbrakk>ip \<in> kD(rt); \<pi>\<^sub>3(r) = unk\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> (\<pi>\<^sub>2(the \<sigma>\<^bsub>route\<^esub>(rt, ip)), \<pi>\<^sub>3(r),
+                                  \<pi>\<^sub>4(r), \<pi>\<^sub>5(r), \<pi>\<^sub>6(r), \<pi>\<^sub>7(addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))))"
+      and c6: "\<lbrakk>ip \<in> kD(rt); sqn rt ip \<ge> \<pi>\<^sub>2(r); \<pi>\<^sub>3(r) = kno;
+                sqn rt ip = \<pi>\<^sub>2(r) \<Longrightarrow> the (dhops rt ip) \<le> \<pi>\<^sub>5(r) \<and> the (flag rt ip) = val\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre (the \<sigma>\<^bsub>route\<^esub>(rt, ip)) (\<pi>\<^sub>7(r))))"
+  shows "(P (update rt ip r))"
+  proof (cases "ip \<in> kD(rt)")
+    assume "ip \<notin> kD(rt)"
+    with c1 show ?thesis
+      by simp
+  next
+    assume "ip \<in> kD(rt)"
+    moreover then obtain dsn dsk fl hops nhip pre
+      where rteq: "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)
+    moreover obtain dsn' dsk' fl' hops' nhip' pre'
+      where req: "r = (dsn', dsk', fl', hops', nhip', pre')"
+        by (cases r) metis
+    ultimately show ?thesis
+      using \<open>(\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)\<close>
+            c2 [OF \<open>ip\<in>kD(rt)\<close>]
+            c3 [OF \<open>ip\<in>kD(rt)\<close>]
+            c4 [OF \<open>ip\<in>kD(rt)\<close>]
+            c5 [OF \<open>ip\<in>kD(rt)\<close>]
+            c6 [OF \<open>ip\<in>kD(rt)\<close>]
+    unfolding update_def sqn_def by auto
+  qed
+
+lemma update_cases_kD:
+  assumes "(\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)"
+      and "ip \<in> kD(rt)"
+      and c2: "sqn rt ip < \<pi>\<^sub>2(r) \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c3: "\<lbrakk>sqn rt ip = \<pi>\<^sub>2(r); the (dhops rt ip) > \<pi>\<^sub>5(r)\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c4: "\<lbrakk>sqn rt ip = \<pi>\<^sub>2(r); the (flag rt ip) = inv\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c5: "\<pi>\<^sub>3(r) = unk \<Longrightarrow> P (rt (ip \<mapsto> (\<pi>\<^sub>2(the \<sigma>\<^bsub>route\<^esub>(rt, ip)), \<pi>\<^sub>3(r),
+                                            \<pi>\<^sub>4(r), \<pi>\<^sub>5(r), \<pi>\<^sub>6(r),
+                                            \<pi>\<^sub>7(addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))))"
+      and c6: "\<lbrakk>sqn rt ip \<ge> \<pi>\<^sub>2(r); \<pi>\<^sub>3(r) = kno;
+                sqn rt ip = \<pi>\<^sub>2(r) \<Longrightarrow> the (dhops rt ip) \<le> \<pi>\<^sub>5(r) \<and> the (flag rt ip) = val\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre (the \<sigma>\<^bsub>route\<^esub>(rt, ip)) (\<pi>\<^sub>7(r))))"
+  shows "(P (update rt ip r))"
+  using assms(1) proof (rule update_cases)
+    assume "sqn rt ip < \<pi>\<^sub>2(r)"
+    thus "P (rt(ip \<mapsto> addpre r (\<pi>\<^sub>7(the (rt ip)))))" by (rule c2)
+  next
+    assume "sqn rt ip = \<pi>\<^sub>2(r)"
+       and "the (dhops rt ip) > \<pi>\<^sub>5(r)"
+    thus "P (rt(ip \<mapsto> addpre r (\<pi>\<^sub>7 (the (rt ip)))))"
+      by (rule c3)
+  next
+    assume "sqn rt ip = \<pi>\<^sub>2(r)"
+       and "the (flag rt ip) = inv"
+    thus "P (rt(ip \<mapsto> addpre r (\<pi>\<^sub>7 (the (rt ip)))))"
+      by (rule c4)
+  next
+    assume "\<pi>\<^sub>3(r) = unk"
+    thus "P (rt (ip \<mapsto> (\<pi>\<^sub>2(the \<sigma>\<^bsub>route\<^esub>(rt, ip)), \<pi>\<^sub>3(r), \<pi>\<^sub>4(r), \<pi>\<^sub>5(r), \<pi>\<^sub>6(r),
+                        \<pi>\<^sub>7(addpre r (\<pi>\<^sub>7(the (rt ip)))))))"
+      by (rule c5)
+  next
+    assume "sqn rt ip \<ge> \<pi>\<^sub>2(r)"
+       and "\<pi>\<^sub>3(r) = kno"
+       and "sqn rt ip = \<pi>\<^sub>2(r) \<Longrightarrow> the (dhops rt ip) \<le> \<pi>\<^sub>5(r) \<and> the (flag rt ip) = val"
+    thus "P (rt (ip \<mapsto> addpre (the (rt ip)) (\<pi>\<^sub>7(r))))"
+      by (rule c6)
+  qed (simp add: \<open>ip \<in> kD(rt)\<close>)
+
+lemma in_kD_after_update [simp]:
+  fixes rt nip dsn dsk flag hops nhip pre
+  shows "kD (update rt nip (dsn, dsk, flag, hops, nhip, pre)) = insert nip (kD rt)"
+  unfolding update_def
+  by (cases "rt nip") auto
+
+lemma nhop_of_update [simp]:
+  fixes rt dip dsn dsk flag hops nhip
+  assumes "rt \<noteq> update rt dip (dsn, dsk, flag, hops, nhip, {})"
+  shows "the (nhop (update rt dip (dsn, dsk, flag, hops, nhip, {})) dip) = nhip"
+  proof -
+  from assms
+  have update_neq: "\<And>v. rt dip = Some v \<Longrightarrow>
+          update rt dip (dsn, dsk, flag, hops, nhip, {})
+             \<noteq> rt(dip \<mapsto> addpre (the (rt dip)) (\<pi>\<^sub>7 (dsn, dsk, flag, hops, nhip, {})))"
+    by auto
+  show ?thesis
+    proof (cases "rt dip = None")
+      assume "rt dip = None"
+      thus "?thesis" unfolding update_def by clarsimp
+    next
+      assume "rt dip \<noteq> None"
+      then obtain v where "rt dip = Some v" by (metis not_None_eq)
+      with update_neq [OF this] show ?thesis
+        unfolding update_def by auto
+    qed
+  qed
+
+lemma sqn_if_updated:
+  fixes rip v rt ip
+  shows "sqn (\<lambda>x. if x = rip then Some v else rt x) ip
+         = (if ip = rip then \<pi>\<^sub>2(v) else sqn rt ip)"
+  unfolding sqn_def by simp
+
+lemma update_sqn [simp]:
+  fixes rt dip rip dsn dsk hops nhip pre
+  assumes "(dsn = 0) = (dsk = unk)"
+  shows "sqn rt dip \<le> sqn (update rt rip (dsn, dsk, val, hops, nhip, pre)) dip"
+  proof (rule update_cases)
+    show "(\<pi>\<^sub>2 (dsn, dsk, val, hops, nhip, pre) = 0) = (\<pi>\<^sub>3 (dsn, dsk, val, hops, nhip, pre) = unk)"
+      by simp (rule assms)
+  qed (clarsimp simp: sqn_if_updated sqn_def)+
+
+lemma sqn_update_bigger [simp]:
+    fixes rt ip ip' dsn dsk flag hops nhip pre
+  assumes "1 \<le> hops"
+    shows "sqn rt ip \<le> sqn (update rt ip' (dsn, dsk, flag, hops, nhip, pre)) ip"
+  using assms unfolding update_def sqn_def
+  by (clarsimp split: option.split) auto
+
+lemma dhops_update [intro]:
+    fixes rt dsn dsk flag hops ip rip nhip pre
+  assumes ex: "\<forall>ip\<in>kD rt. the (dhops rt ip) \<ge> 1"
+      and ip: "(ip = rip \<and> Suc 0 \<le> hops) \<or> (ip \<noteq> rip \<and> ip\<in>kD rt)"
+    shows "Suc 0 \<le> the (dhops (update rt rip (dsn, dsk, flag, hops, nhip, pre)) ip)"
+  using ip proof
+    assume "ip = rip \<and> Suc 0 \<le> hops" thus ?thesis
+      unfolding update_def using ex
+      by (cases "rip \<in> kD rt") (drule(1) bspec, auto)
+  next
+    assume "ip \<noteq> rip \<and> ip\<in>kD rt" thus ?thesis
+      using ex unfolding update_def
+      by (cases "rip\<in>kD rt") auto
+  qed
+
+lemma update_another [simp]:
+    fixes dip ip rt dsn dsk flag hops nhip pre
+  assumes "ip \<noteq> dip"
+    shows "(update rt dip (dsn, dsk, flag, hops, nhip, pre)) ip = rt ip"
+  using assms unfolding update_def
+  by (clarsimp split: option.split)
+
+lemma nhop_update_another [simp]:
+    fixes dip ip rt dsn dsk flag hops nhip pre
+  assumes "ip \<noteq> dip"
+    shows "nhop (update rt dip (dsn, dsk, flag, hops, nhip, pre)) ip = nhop rt ip"
+  using assms unfolding update_def
+  by (clarsimp split: option.split)
+
+lemma dhops_update_another [simp]:
+    fixes dip ip rt dsn dsk flag hops nhip pre
+  assumes "ip \<noteq> dip"
+    shows "dhops (update rt dip (dsn, dsk, flag, hops, nhip, pre)) ip = dhops rt ip"
+  using assms unfolding update_def
+  by (clarsimp split: option.split)
+
+lemma sqn_update_same [simp]:
+  "\<And>rt ip dsn dsk flag hops nhip pre. sqn (rt(ip \<mapsto> v)) ip = \<pi>\<^sub>2(v)"
+  unfolding sqn_def by simp
+
+lemma dhops_update_changed [simp]:
+    fixes rt dip osn hops nhip
+  assumes "rt \<noteq> update rt dip (osn, kno, val, hops, nhip, {})"
+    shows "the (dhops (update rt dip (osn, kno, val, hops, nhip, {})) dip) = hops"
+  using assms unfolding update_def                                                      
+  by (clarsimp split: option.split_asm option.split if_split_asm) auto
+
+lemma nhop_update_unk_val [simp]:
+  "\<And>rt dip ip dsn hops npre.
+   the (nhop (update rt dip (dsn, unk, val, hops, ip, npre)) dip) = ip"
+   unfolding update_def by (clarsimp split: option.split)
+
+lemma nhop_update_changed [simp]:
+    fixes rt dip dsn dsk flg hops sip
+  assumes "update rt dip (dsn, dsk, flg, hops, sip, {}) \<noteq> rt"
+    shows "the (nhop (update rt dip (dsn, dsk, flg, hops, sip, {})) dip) = sip"
+  using assms unfolding update_def
+  by (clarsimp split: option.splits if_split_asm) auto
+
+lemma update_rt_split_asm:
+  "\<And>rt ip dsn dsk flag hops sip.
+   P (update rt ip (dsn, dsk, flag, hops, sip, {}))
+   =
+   (\<not>(rt = update rt ip (dsn, dsk, flag, hops, sip, {}) \<and> \<not>P rt
+      \<or> rt \<noteq> update rt ip (dsn, dsk, flag, hops, sip, {})
+         \<and> \<not>P (update rt ip (dsn, dsk, flag, hops, sip, {}))))"
+  by auto
+
+lemma sqn_update [simp]: "\<And>rt dip dsn flg hops sip.
+  rt \<noteq> update rt dip (dsn, kno, flg, hops, sip, {})
+  \<Longrightarrow> sqn (update rt dip (dsn, kno, flg, hops, sip, {})) dip = dsn"
+  unfolding update_def by (clarsimp split: option.split if_split_asm) auto
+
+lemma sqnf_update [simp]: "\<And>rt dip dsn dsk flg hops sip.
+  rt \<noteq> update rt dip (dsn, dsk, flg, hops, sip, {})
+  \<Longrightarrow> sqnf (update rt dip (dsn, dsk, flg, hops, sip, {})) dip = dsk"
+  unfolding update_def sqnf_def
+  by (clarsimp split: option.splits if_split_asm) auto
+
+lemma update_kno_dsn_greater_zero:
+  "\<And>rt dip ip dsn hops npre. 1 \<le> dsn \<Longrightarrow> 1 \<le> (sqn (update rt dip (dsn, kno, val, hops, ip, npre)) dip)"
+   unfolding update_def 
+  by (clarsimp split: option.splits)
+
+lemma proj3_update [simp]: "\<And>rt dip dsn dsk flg hops sip.
+  rt \<noteq> update rt dip (dsn, dsk, flg, hops, sip, {})
+  \<Longrightarrow> \<pi>\<^sub>3(the (update rt dip (dsn, dsk, flg, hops, sip, {}) dip)) = dsk"
+  unfolding update_def sqnf_def
+  by (clarsimp split: option.splits if_split_asm) auto
+
+lemma nhop_update_changed_kno_val [simp]: "\<And>rt ip dsn dsk hops nhip.
+  rt \<noteq> update rt ip (dsn, kno, val, hops, nhip, {})
+   \<Longrightarrow> the (nhop (update rt ip (dsn, kno, val, hops, nhip, {})) ip) = nhip"
+  unfolding update_def
+  by (clarsimp split: option.split_asm option.split if_split_asm) auto
+
+lemma flag_update [simp]: "\<And>rt dip dsn flg hops sip.
+  rt \<noteq> update rt dip (dsn, kno, flg, hops, sip, {})
+  \<Longrightarrow> the (flag (update rt dip (dsn, kno, flg, hops, sip, {})) dip) = flg"
+  unfolding update_def
+  by (clarsimp split: option.split if_split_asm) auto
+
+lemma the_flag_Some [dest!]:
+    fixes ip rt
+  assumes "the (flag rt ip) = x"
+      and "ip \<in> kD rt"
+    shows "flag rt ip = Some x"
+  using assms by auto
+
+lemma kD_update_unchanged [dest]:
+    fixes rt dip dsn dsk flag hops nhip pre
+  assumes "rt = update rt dip (dsn, dsk, flag, hops, nhip, pre)"
+    shows "dip\<in>kD(rt)"
+  proof -
+    have "dip\<in>kD(update rt dip (dsn, dsk, flag, hops, nhip, pre))" by simp
+    with assms show ?thesis by simp
+  qed
+
+lemma nhop_update [simp]: "\<And>rt dip dsn dsk flg hops sip.
+  rt \<noteq> update rt dip (dsn, dsk, flg, hops, sip, {})
+  \<Longrightarrow> the (nhop (update rt dip (dsn, dsk, flg, hops, sip, {})) dip) = sip"
+  unfolding update_def sqnf_def
+  by (clarsimp split: option.splits if_split_asm) auto
+
+lemma sqn_update_another [simp]:
+    fixes dip ip rt dsn dsk flag hops nhip pre
+  assumes "ip \<noteq> dip"
+    shows "sqn (update rt dip (dsn, dsk, flag, hops, nhip, pre)) ip = sqn rt ip"
+  using assms unfolding update_def sqn_def
+  by (clarsimp split: option.splits) auto
+
+lemma sqnf_update_another [simp]:
+    fixes dip ip rt dsn dsk flag hops nhip pre
+  assumes "ip \<noteq> dip"
+    shows "sqnf (update rt dip (dsn, dsk, flag, hops, nhip, pre)) ip = sqnf rt ip"
+  using assms unfolding update_def sqnf_def
+  by (clarsimp split: option.splits) auto
+
+lemma vD_update_val [dest]:
+  "\<And>dip rt dip' dsn dsk hops nhip pre.
+   dip \<in> vD(update rt dip' (dsn, dsk, val, hops, nhip, pre)) \<Longrightarrow> (dip\<in>vD(rt) \<or> dip=dip')"
+   unfolding update_def vD_def by (clarsimp split: option.split_asm if_split_asm)
+
+subsubsection "Invalidating route entries"
+
+definition invalidate :: "rt \<Rightarrow> (ip \<rightharpoonup> sqn) \<Rightarrow> rt"
+where "invalidate rt dests \<equiv>
+  \<lambda>ip. case (rt ip, dests ip) of
+    (None, _) \<Rightarrow> None
+  | (Some s, None) \<Rightarrow> Some s
+  | (Some (_, dsk, _, hops, nhip, pre), Some rsn) \<Rightarrow>
+                      Some (rsn, dsk, inv, hops, nhip, pre)"
+
+lemma proj3_invalidate [simp]:
+  "\<And>dip. \<pi>\<^sub>3(the ((invalidate rt dests) dip)) = \<pi>\<^sub>3(the (rt dip))"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+lemma proj5_invalidate [simp]:
+  "\<And>dip. \<pi>\<^sub>5(the ((invalidate rt dests) dip)) = \<pi>\<^sub>5(the (rt dip))"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+lemma proj6_invalidate [simp]:
+  "\<And>dip. \<pi>\<^sub>6(the ((invalidate rt dests) dip)) = \<pi>\<^sub>6(the (rt dip))"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+lemma proj7_invalidate [simp]:
+  "\<And>dip. \<pi>\<^sub>7(the ((invalidate rt dests) dip)) = \<pi>\<^sub>7(the (rt dip))"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+subsection "Route Requests"
+
+lemma invalidate_kD_inv [simp]:
+  "\<And>rt dests. kD (invalidate rt dests) = kD rt"
+  unfolding invalidate_def kD_def
+  by (simp split: option.split)
+
+lemma invalidate_sqn:
+  fixes rt dip dests
+  assumes "\<forall>rsn. dests dip = Some rsn \<longrightarrow> sqn rt dip \<le> rsn"
+  shows "sqn rt dip \<le> sqn (invalidate rt dests) dip"
+  proof (cases "dip \<notin> kD(rt)")
+    assume "\<not> dip \<notin> kD(rt)"
+    hence "dip\<in>kD(rt)" by simp
+    then obtain dsn dsk flag hops nhip pre where "rt dip = Some (dsn, dsk, flag, hops, nhip, pre)"
+      by (metis kD_Some)
+    with assms show "sqn rt dip \<le> sqn (invalidate rt dests) dip"
+      by (cases "dests dip") (auto simp add: invalidate_def sqn_def)
+  qed simp
+
+lemma sqn_invalidate_in_dests [simp]:
+    fixes dests ipa rsn rt
+  assumes "dests ipa = Some rsn"
+      and "ipa\<in>kD(rt)"
+    shows "sqn (invalidate rt dests) ipa = rsn"
+  unfolding invalidate_def sqn_def
+  using assms(1) assms(2) [THEN kD_Some]
+  by clarsimp
+
+lemma dhops_invalidate [simp]:
+  "\<And>dip. the (dhops (invalidate rt dests) dip) = the (dhops rt dip)"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+lemma sqnf_invalidate [simp]:
+  "\<And>dip. sqnf (invalidate (rt \<xi>) (dests \<xi>)) dip = sqnf (rt \<xi>) dip"
+  unfolding sqnf_def invalidate_def by (clarsimp split: option.split)
+
+lemma nhop_invalidate [simp]:
+  "\<And>dip. the (nhop (invalidate (rt \<xi>) (dests \<xi>)) dip) = the (nhop (rt \<xi>) dip)"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+lemma invalidate_other [simp]:
+    fixes rt dests dip
+  assumes "dip\<notin>dom(dests)"
+    shows "invalidate rt dests dip = rt dip"
+  using assms unfolding invalidate_def
+  by (clarsimp split: option.split_asm)
+
+lemma invalidate_none [simp]:
+    fixes rt dests dip
+  assumes "dip\<notin>kD(rt)"
+    shows "invalidate rt dests dip = None"
+  using assms unfolding invalidate_def by clarsimp
+
+lemma vD_invalidate_vD_not_dests:
+  "\<And>dip rt dests. dip\<in>vD(invalidate rt dests) \<Longrightarrow> dip\<in>vD(rt) \<and> dests dip = None"
+  unfolding invalidate_def vD_def 
+  by (clarsimp split: option.split_asm)
+
+lemma sqn_invalidate_not_in_dests [simp]:
+  fixes dests dip rt
+  assumes "dip\<notin>dom(dests)"
+  shows "sqn (invalidate rt dests) dip = sqn rt dip"
+  using assms unfolding sqn_def by simp
+
+lemma invalidate_changes:
+    fixes rt dests dip dsn dsk flag hops nhip pre
+  assumes "invalidate rt dests dip = Some (dsn, dsk, flag, hops, nhip, pre)"
+    shows "  dsn = (case dests dip of None \<Rightarrow> \<pi>\<^sub>2(the (rt dip)) | Some rsn \<Rightarrow> rsn)
+           \<and> dsk = \<pi>\<^sub>3(the (rt dip))
+           \<and> flag = (if dests dip = None then \<pi>\<^sub>4(the (rt dip)) else inv)
+           \<and> hops = \<pi>\<^sub>5(the (rt dip))
+           \<and> nhip = \<pi>\<^sub>6(the (rt dip))
+           \<and> pre = \<pi>\<^sub>7(the (rt dip))"
+  using assms unfolding invalidate_def
+  by (cases "rt dip", clarsimp, cases "dests dip") auto
+
+
+lemma proj3_inv: "\<And>dip rt dests. dip\<in>kD (rt)
+                      \<Longrightarrow> \<pi>\<^sub>3(the (invalidate rt dests dip)) = \<pi>\<^sub>3(the (rt dip))"
+  by (clarsimp simp: invalidate_def kD_def split: option.split)
+
+lemma dests_iD_invalidate [simp]:
+  assumes "dests ip = Some rsn"
+      and "ip\<in>kD(rt)"
+    shows "ip\<in>iD(invalidate rt dests)"
+  using assms(1) assms(2) [THEN kD_Some] unfolding invalidate_def iD_def
+  by (clarsimp split: option.split)
+
+subsection "Queued Packets"
+
+text \<open>Functions for sending data packets.\<close>
+
+type_synonym store = "ip \<rightharpoonup> (p \<times> data list)"
+
+definition sigma_queue :: "store \<Rightarrow> ip \<Rightarrow> data list"    ("\<sigma>\<^bsub>queue\<^esub>'(_, _')")
+  where "\<sigma>\<^bsub>queue\<^esub>(store, dip) \<equiv> case store dip of None \<Rightarrow> [] | Some (p, q) \<Rightarrow> q"
+
+definition qD :: "store \<Rightarrow> ip set"
+  where "qD \<equiv> dom"
+
+definition add :: "data \<Rightarrow> ip \<Rightarrow> store \<Rightarrow> store"
+  where "add d dip store \<equiv> case store dip of
+                              None \<Rightarrow> store (dip \<mapsto> (req, [d]))
+                            | Some (p, q) \<Rightarrow> store (dip \<mapsto> (p, q @ [d]))"
+
+lemma qD_add [simp]:
+  fixes d dip store
+  shows "qD(add d dip store) = insert dip (qD store)"
+  unfolding add_def Let_def qD_def
+  by (clarsimp split: option.split)
+
+definition drop :: "ip \<Rightarrow> store \<rightharpoonup> store"
+  where "drop dip store \<equiv>
+    map_option (\<lambda>(p, q). if tl q = [] then store (dip := None)
+                                      else store (dip \<mapsto> (p, tl q))) (store dip)"
+
+definition sigma_p_flag :: "store \<Rightarrow> ip \<rightharpoonup> p" ("\<sigma>\<^bsub>p-flag\<^esub>'(_, _')")
+  where "\<sigma>\<^bsub>p-flag\<^esub>(store, dip) \<equiv> map_option fst (store dip)"
+
+definition unsetRRF :: "store \<Rightarrow> ip \<Rightarrow> store"
+  where "unsetRRF store dip \<equiv> case store dip of
+                                None \<Rightarrow> store
+                              | Some (p, q) \<Rightarrow> store (dip \<mapsto> (noreq, q))"
+
+definition setRRF :: "store \<Rightarrow> (ip \<rightharpoonup> sqn) \<Rightarrow> store"
+  where "setRRF store dests \<equiv> \<lambda>dip. if dests dip = None then store dip
+                                    else map_option (\<lambda>(_, q). (req, q)) (store dip)"
+
+subsection "Comparison with the original technical report"
+
+text \<open>
+  The major differences with the AODV technical report of Fehnker et al are:
+  \begin{enumerate}
+  \item @{term nhop} is partial, thus a `@{term the}' is needed, similarly for @{term dhops}
+        and @{term addpreRT}.
+  \item @{term precs} is partial.
+  \item @{term "\<sigma>\<^bsub>p-flag\<^esub>(store, dip)"} is partial.
+  \item The routing table (@{typ rt}) is modelled as a map (@{typ "ip \<Rightarrow> r option"})
+        rather than a set of 7-tuples, likewise, the @{typ r} is a 6-tuple rather than
+        a 7-tuple, i.e., the destination ip-address (@{term "dip"}) is taken from the
+        argument to the function, rather than a part of the result. Well-definedness then
+        follows from the structure of the type and more related facts are available
+        automatically, rather than having to be acquired through tedious proofs.
+  \item Similar remarks hold for the dests mapping passed to @{term "invalidate"},
+        and @{term "store"}.
+  \end{enumerate}
+\<close>
+
+end
+

formal/afp/AODV/variants/a_norreqid/A_Aodv_Loop_Freedom.thy

@@ -0,0 +1,369 @@
+(*  Title:       variants/a_norreqid/Aodv_Loop_Freedom.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "Lift and transfer invariants to show loop freedom"
+
+theory A_Aodv_Loop_Freedom
+imports AWN.OClosed_Transfer AWN.Qmsg_Lifting A_Global_Invariants A_Loop_Freedom
+begin
+
+text \<open>lift to parallel processes with queues\<close>
+
+lemma par_step_no_change_on_send_or_receive:
+    fixes \<sigma> s a \<sigma>' s'
+  assumes "((\<sigma>, s), a, (\<sigma>', s')) \<in> oparp_sos i (oseqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i) (seqp_sos \<Gamma>\<^sub>Q\<^sub>M\<^sub>S\<^sub>G)"
+      and "a \<noteq> \<tau>"
+    shows "\<sigma>' i = \<sigma> i"
+  using assms by (rule qmsg_no_change_on_send_or_receive)
+
+lemma par_nhop_quality_increases:
+  shows "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile> (otherwith ((=)) {i} (orecvmsg (\<lambda>\<sigma> m.
+                                    msg_fresh \<sigma> m \<and> msg_zhops m)),
+                                  other quality_increases {i} \<rightarrow>)
+                        global (\<lambda>\<sigma>. \<forall>dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                          in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                                             \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+    proof (rule lift_into_qmsg [OF seq_nhop_quality_increases])
+    show "opaodv i \<Turnstile>\<^sub>A (otherwith ((=)) {i}
+                         (orecvmsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)),
+                        other quality_increases {i} \<rightarrow>)
+                       globala (\<lambda>(\<sigma>, _, \<sigma>'). quality_increases (\<sigma> i) (\<sigma>' i))"
+    proof (rule ostep_invariant_weakenE [OF oquality_increases], simp_all)
+      fix t :: "(((nat \<Rightarrow> state) \<times> (state, msg, pseqp, pseqp label) seqp), msg seq_action) transition"
+      assume "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<sigma>, _), _, (\<sigma>', _)). \<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)) t"
+      thus "quality_increases (fst (fst t) i) (fst (snd (snd t)) i)"
+        by (cases t) (clarsimp dest!: onllD, metis aodv_ex_label)
+    next
+      fix \<sigma> \<sigma>' a
+      assume "otherwith ((=)) {i}
+                (orecvmsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)) \<sigma> \<sigma>' a"
+      thus "otherwith quality_increases {i} (orecvmsg (\<lambda>_. rreq_rrep_sn)) \<sigma> \<sigma>' a"
+        by - (erule weaken_otherwith, auto)
+    qed
+  qed auto
+
+lemma par_rreq_rrep_sn_quality_increases:
+  "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                            globala (\<lambda>(\<sigma>, _, \<sigma>'). quality_increases (\<sigma> i) (\<sigma>' i))"
+  proof -
+    have "opaodv i \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                globala (\<lambda>(\<sigma>, _, \<sigma>'). quality_increases (\<sigma> i) (\<sigma>' i))"
+      by (rule ostep_invariant_weakenE [OF olocal_quality_increases])
+         (auto dest!: onllD seqllD elim!: aodv_ex_labelE)
+    hence "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                                   globala (\<lambda>(\<sigma>, _, \<sigma>'). quality_increases (\<sigma> i) (\<sigma>' i))"
+      by (rule lift_step_into_qmsg_statelessassm) simp_all
+    thus ?thesis by rule auto
+  qed
+
+lemma par_rreq_rrep_nsqn_fresh_any_step:
+  shows "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>,
+                                   other (\<lambda>_ _. True) {i} \<rightarrow>)
+                                  globala (\<lambda>(\<sigma>, a, \<sigma>'). anycast (msg_fresh \<sigma>) a)"
+  proof -
+    have "opaodv i \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. (orecvmsg (\<lambda>_. rreq_rrep_sn)) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                       globala (\<lambda>(\<sigma>, a, \<sigma>'). anycast (msg_fresh \<sigma>) a)"
+    proof (rule ostep_invariant_weakenE [OF rreq_rrep_nsqn_fresh_any_step_invariant])
+      fix t
+      assume "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<sigma>, _), a, _). anycast (msg_fresh \<sigma>) a) t"
+      thus "globala (\<lambda>(\<sigma>, a, \<sigma>'). anycast (msg_fresh \<sigma>) a) t"
+        by (cases t) (clarsimp dest!: onllD, metis aodv_ex_label)
+    qed auto
+    hence "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. (orecvmsg (\<lambda>_. rreq_rrep_sn)) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                                    globala (\<lambda>(\<sigma>, a, \<sigma>'). anycast (msg_fresh \<sigma>) a)"
+      by (rule lift_step_into_qmsg_statelessassm) simp_all
+    thus ?thesis by rule auto
+  qed
+
+lemma par_anycast_msg_zhops:
+  shows "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                                  globala (\<lambda>(_, a, _). anycast msg_zhops a)"
+  proof -
+    from anycast_msg_zhops initiali_aodv oaodv_trans aodv_trans
+      have "opaodv i \<Turnstile>\<^sub>A (act TT, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                         seqll i (onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(_, a, _). anycast msg_zhops a))"
+        by (rule open_seq_step_invariant)
+    hence "opaodv i \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                            globala (\<lambda>(_, a, _). anycast msg_zhops a)"
+    proof (rule ostep_invariant_weakenE)
+      fix t :: "(((nat \<Rightarrow> state) \<times> (state, msg, pseqp, pseqp label) seqp), msg seq_action) transition"
+      assume "seqll i (onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(_, a, _). anycast msg_zhops a)) t"
+      thus "globala (\<lambda>(_, a, _). anycast msg_zhops a) t"
+        by (cases t) (clarsimp dest!: seqllD onllD, metis aodv_ex_label)
+    qed simp_all
+    hence "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                                     globala (\<lambda>(_, a, _). anycast msg_zhops a)"
+      by (rule lift_step_into_qmsg_statelessassm) simp_all
+    thus ?thesis by rule auto
+  qed
+
+subsection \<open>Lift to nodes\<close>
+
+lemma node_step_no_change_on_send_or_receive:
+  assumes "((\<sigma>, NodeS i P R), a, (\<sigma>', NodeS i' P' R')) \<in> onode_sos
+                                      (oparp_sos i (oseqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i) (seqp_sos \<Gamma>\<^sub>Q\<^sub>M\<^sub>S\<^sub>G))"
+      and "a \<noteq> \<tau>"
+    shows "\<sigma>' i = \<sigma> i"
+  using assms
+  by (cases a) (auto elim!: par_step_no_change_on_send_or_receive)
+
+lemma node_nhop_quality_increases:
+  shows "\<langle> i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R \<rangle>\<^sub>o \<Turnstile>
+           (otherwith ((=)) {i}
+              (oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)),
+              other quality_increases {i}
+            \<rightarrow>) global (\<lambda>\<sigma>. \<forall>dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                  in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                                     \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+  by (rule node_lift [OF par_nhop_quality_increases]) auto
+
+lemma node_quality_increases:
+  "\<langle> i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R \<rangle>\<^sub>o \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma>,
+                                         other (\<lambda>_ _. True) {i} \<rightarrow>)
+                            globala (\<lambda>(\<sigma>, _, \<sigma>'). quality_increases (\<sigma> i) (\<sigma>' i))"
+  by (rule node_lift_step_statelessassm [OF par_rreq_rrep_sn_quality_increases]) simp
+
+lemma node_rreq_rrep_nsqn_fresh_any_step:
+  shows "\<langle> i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R \<rangle>\<^sub>o \<Turnstile>\<^sub>A
+          (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+          globala (\<lambda>(\<sigma>, a, \<sigma>'). castmsg (msg_fresh \<sigma>) a)"
+  by (rule node_lift_anycast_statelessassm [OF par_rreq_rrep_nsqn_fresh_any_step])
+
+lemma node_anycast_msg_zhops:
+  shows "\<langle> i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R \<rangle>\<^sub>o \<Turnstile>\<^sub>A
+          (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+          globala (\<lambda>(_, a, _). castmsg msg_zhops a)"
+  by (rule node_lift_anycast_statelessassm [OF par_anycast_msg_zhops])
+
+lemma node_silent_change_only:
+  shows "\<langle> i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<^sub>i \<rangle>\<^sub>o \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_ _. True) \<sigma>,
+                                               other (\<lambda>_ _. True) {i} \<rightarrow>)
+          globala (\<lambda>(\<sigma>, a, \<sigma>'). a \<noteq> \<tau> \<longrightarrow> \<sigma>' i = \<sigma> i)"
+  proof (rule ostep_invariantI, simp (no_asm), rule impI)
+    fix \<sigma> \<zeta> a \<sigma>' \<zeta>'
+    assume or: "(\<sigma>, \<zeta>) \<in> oreachable (\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<^sub>i\<rangle>\<^sub>o)
+                                    (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_ _. True) \<sigma>)
+                                    (other (\<lambda>_ _. True) {i})"
+      and tr: "((\<sigma>, \<zeta>), a, (\<sigma>', \<zeta>')) \<in> trans (\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<^sub>i\<rangle>\<^sub>o)"
+      and "a \<noteq> \<tau>\<^sub>n"
+    from or obtain p R where "\<zeta> = NodeS i p R"
+      by - (drule node_net_state, metis)
+    with tr have "((\<sigma>, NodeS i p R), a, (\<sigma>', \<zeta>'))
+                     \<in> onode_sos (oparp_sos i (trans (opaodv i)) (trans qmsg))"
+      by simp
+    thus "\<sigma>' i = \<sigma> i" using \<open>a \<noteq> \<tau>\<^sub>n\<close> 
+      by (cases rule: onode_sos.cases)
+         (auto elim: qmsg_no_change_on_send_or_receive)
+  qed
+
+subsection \<open>Lift to partial networks\<close>
+
+lemma arrive_rreq_rrep_nsqn_fresh_inc_sn [simp]:
+  assumes "oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> P \<sigma> m) \<sigma> m"
+    shows "oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma> m"
+  using assms by (cases m) auto
+
+lemma opnet_nhop_quality_increases:
+  shows "opnet (\<lambda>i. opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg) p \<Turnstile>
+           (otherwith ((=)) (net_tree_ips p)
+              (oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)),
+               other quality_increases (net_tree_ips p) \<rightarrow>)
+              global (\<lambda>\<sigma>. \<forall>i\<in>net_tree_ips p. \<forall>dip.
+                          let nhip = the (nhop (rt (\<sigma> i)) dip)
+                          in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                             \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+  proof (rule pnet_lift [OF node_nhop_quality_increases])
+    fix i R
+    have "\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma>,
+                                              other (\<lambda>_ _. True) {i} \<rightarrow>) globala (\<lambda>(\<sigma>, a, \<sigma>').
+            castmsg (\<lambda>m. msg_fresh \<sigma> m \<and> msg_zhops m) a)"
+    proof (rule ostep_invariantI, simp (no_asm))
+      fix \<sigma> s a \<sigma>' s'
+      assume or: "(\<sigma>, s) \<in> oreachable (\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o)
+                                      (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma>)
+                                      (other (\<lambda>_ _. True) {i})"
+         and tr: "((\<sigma>, s), a, (\<sigma>', s')) \<in> trans (\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o)"
+         and am: "oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma> a"
+      from or tr am have "castmsg (msg_fresh \<sigma>) a"
+        by (auto dest!: ostep_invariantD [OF node_rreq_rrep_nsqn_fresh_any_step])
+      moreover from or tr am have "castmsg (msg_zhops) a"
+        by (auto dest!: ostep_invariantD [OF node_anycast_msg_zhops])
+      ultimately show "castmsg (\<lambda>m. msg_fresh \<sigma> m \<and> msg_zhops m) a"
+        by (case_tac a) auto
+    qed
+    thus "\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o \<Turnstile>\<^sub>A
+            (\<lambda>\<sigma> _. oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m) \<sigma>,
+             other quality_increases {i} \<rightarrow>) globala (\<lambda>(\<sigma>, a, _).
+               castmsg (\<lambda>m. msg_fresh \<sigma> m \<and> msg_zhops m) a)"
+      by rule auto
+  next
+    fix i R
+    show "\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o \<Turnstile>\<^sub>A
+            (\<lambda>\<sigma> _. oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m) \<sigma>,
+             other quality_increases {i} \<rightarrow>) globala (\<lambda>(\<sigma>, a, \<sigma>').
+               a \<noteq> \<tau> \<and> (\<forall>d. a \<noteq> i:deliver(d)) \<longrightarrow> \<sigma> i = \<sigma>' i)"
+      by (rule ostep_invariant_weakenE [OF node_silent_change_only]) auto
+  next
+    fix i R
+    show "\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o \<Turnstile>\<^sub>A
+            (\<lambda>\<sigma> _. oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m) \<sigma>,
+             other quality_increases {i} \<rightarrow>) globala (\<lambda>(\<sigma>, a, \<sigma>').
+               a = \<tau> \<or> (\<exists>d. a = i:deliver(d)) \<longrightarrow> quality_increases (\<sigma> i) (\<sigma>' i))"
+      by (rule ostep_invariant_weakenE [OF node_quality_increases]) auto
+  qed simp_all
+
+subsection \<open>Lift to closed networks\<close>
+
+lemma onet_nhop_quality_increases:
+  shows "oclosed (opnet (\<lambda>i. opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg) p)
+           \<Turnstile> (\<lambda>_ _ _. True, other quality_increases (net_tree_ips p) \<rightarrow>)
+              global (\<lambda>\<sigma>. \<forall>i\<in>net_tree_ips p. \<forall>dip.
+                          let nhip = the (nhop (rt (\<sigma> i)) dip)
+                          in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                             \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+  (is "_ \<Turnstile> (_, ?U \<rightarrow>) ?inv")
+  proof (rule inclosed_closed)
+    from opnet_nhop_quality_increases
+      show "opnet (\<lambda>i. opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg) p
+               \<Turnstile> (otherwith ((=)) (net_tree_ips p) inoclosed, ?U \<rightarrow>) ?inv"
+    proof (rule oinvariant_weakenE)
+      fix \<sigma> \<sigma>' :: "ip \<Rightarrow> state" and a :: "msg node_action"
+      assume "otherwith ((=)) (net_tree_ips p) inoclosed \<sigma> \<sigma>' a"
+      thus "otherwith ((=)) (net_tree_ips p)
+              (oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)) \<sigma> \<sigma>' a"
+      proof (rule otherwithEI)
+        fix \<sigma> :: "ip \<Rightarrow> state" and a :: "msg node_action"
+        assume "inoclosed \<sigma> a"
+        thus "oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m) \<sigma> a"
+        proof (cases a)
+          fix ii ni ms
+          assume "a = ii\<not>ni:arrive(ms)"
+          moreover with \<open>inoclosed \<sigma> a\<close> obtain d di where "ms = newpkt(d, di)"
+            by (cases ms) auto
+          ultimately show ?thesis by simp
+        qed simp_all
+      qed
+    qed
+  qed
+
+subsection \<open>Transfer into the standard model\<close>
+
+interpretation aodv_openproc: openproc paodv opaodv id
+  rewrites "aodv_openproc.initmissing = initmissing"
+  proof -
+    show "openproc paodv opaodv id"
+    proof unfold_locales
+      fix i :: ip
+      have "{(\<sigma>, \<zeta>). (\<sigma> i, \<zeta>) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i \<and> (\<forall>j. j \<noteq> i \<longrightarrow> \<sigma> j \<in> fst ` \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V j)} \<subseteq> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'"
+        unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'_def
+        proof (rule equalityD1)
+          show "\<And>f p. {(\<sigma>, \<zeta>). (\<sigma> i, \<zeta>) \<in> {(f i, p)} \<and> (\<forall>j. j \<noteq> i
+                      \<longrightarrow> \<sigma> j \<in> fst ` {(f j, p)})} = {(f, p)}"
+            by (rule set_eqI) auto
+        qed
+      thus "{ (\<sigma>, \<zeta>) |\<sigma> \<zeta> s. s \<in> init (paodv i)
+                             \<and> (\<sigma> i, \<zeta>) = id s
+                             \<and> (\<forall>j. j\<noteq>i \<longrightarrow> \<sigma> j \<in> (fst o id) ` init (paodv j)) } \<subseteq> init (opaodv i)"
+        by simp
+    next
+      show "\<forall>j. init (paodv j) \<noteq> {}"
+        unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by simp
+    next
+      fix i s a s' \<sigma> \<sigma>'
+      assume "\<sigma> i = fst (id s)"
+         and "\<sigma>' i = fst (id s')"
+         and "(s, a, s') \<in> trans (paodv i)"
+      then obtain q q' where "s = (\<sigma> i, q)"
+                         and "s' = (\<sigma>' i, q')"
+                         and "((\<sigma> i, q), a, (\<sigma>' i, q')) \<in> trans (paodv i)" 
+         by (cases s, cases s') auto
+      from this(3) have "((\<sigma>, q), a, (\<sigma>', q')) \<in> trans (opaodv i)"
+        by simp (rule open_seqp_action [OF aodv_wf])
+
+      with \<open>s = (\<sigma> i, q)\<close> and \<open>s' = (\<sigma>' i, q')\<close>
+        show "((\<sigma>, snd (id s)), a, (\<sigma>', snd (id s'))) \<in> trans (opaodv i)"
+          by simp
+    qed
+    then interpret opn: openproc paodv opaodv id .
+    have [simp]: "\<And>i. (SOME x. x \<in> (fst o id) ` init (paodv i)) = aodv_init i"
+      unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by simp
+    hence "\<And>i. openproc.initmissing paodv id i = initmissing i"
+      unfolding opn.initmissing_def opn.someinit_def initmissing_def
+      by (auto split: option.split)
+    thus "openproc.initmissing paodv id = initmissing" ..
+  qed
+
+interpretation aodv_openproc_par_qmsg: openproc_parq paodv opaodv id qmsg
+  rewrites "aodv_openproc_par_qmsg.netglobal = netglobal"
+    and "aodv_openproc_par_qmsg.initmissing = initmissing"
+  proof -
+    show "openproc_parq paodv opaodv id qmsg"
+      by (unfold_locales) simp
+    then interpret opq: openproc_parq paodv opaodv id qmsg .
+
+    have im: "\<And>\<sigma>. openproc.initmissing (\<lambda>i. paodv i \<langle>\<langle> qmsg) (\<lambda>(p, q). (fst (id p), snd (id p), q)) \<sigma>
+                                                                                    = initmissing \<sigma>"
+      unfolding opq.initmissing_def opq.someinit_def initmissing_def
+      unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def \<sigma>\<^sub>Q\<^sub>M\<^sub>S\<^sub>G_def by (clarsimp cong: option.case_cong)
+    thus "openproc.initmissing (\<lambda>i. paodv i \<langle>\<langle> qmsg) (\<lambda>(p, q). (fst (id p), snd (id p), q)) = initmissing"
+      by (rule ext)
+    have "\<And>P \<sigma>. openproc.netglobal (\<lambda>i. paodv i \<langle>\<langle> qmsg) (\<lambda>(p, q). (fst (id p), snd (id p), q)) P \<sigma>
+                                                                                = netglobal P \<sigma>"
+      unfolding opq.netglobal_def netglobal_def opq.initmissing_def initmissing_def opq.someinit_def
+      unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def \<sigma>\<^sub>Q\<^sub>M\<^sub>S\<^sub>G_def
+      by (clarsimp cong: option.case_cong
+                   simp del: One_nat_def
+                   simp add: fst_initmissing_netgmap_default_aodv_init_netlift
+                                                  [symmetric, unfolded initmissing_def])
+    thus "openproc.netglobal (\<lambda>i. paodv i \<langle>\<langle> qmsg) (\<lambda>(p, q). (fst (id p), snd (id p), q)) = netglobal"
+      by auto
+  qed
+
+lemma net_nhop_quality_increases:
+  assumes "wf_net_tree n"
+  shows "closed (pnet (\<lambda>i. paodv i \<langle>\<langle> qmsg) n) \<TTurnstile> netglobal
+                           (\<lambda>\<sigma>. \<forall>i dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                        in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                                            \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+        (is "_ \<TTurnstile> netglobal (\<lambda>\<sigma>. \<forall>i. ?inv \<sigma> i)")
+  proof -
+    from \<open>wf_net_tree n\<close>
+      have proto: "closed (pnet (\<lambda>i. paodv i \<langle>\<langle> qmsg) n) \<TTurnstile> netglobal (\<lambda>\<sigma>. \<forall>i\<in>net_tree_ips n. \<forall>dip.
+                                            let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                            in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                                                \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+        by (rule aodv_openproc_par_qmsg.close_opnet [OF _ onet_nhop_quality_increases])
+    show ?thesis
+    unfolding invariant_def opnet_sos.opnet_tau1
+    proof (rule, simp only: aodv_openproc_par_qmsg.netglobalsimp
+                            fst_initmissing_netgmap_pair_fst, rule allI)
+      fix \<sigma> i
+      assume sr: "\<sigma> \<in> reachable (closed (pnet (\<lambda>i. paodv i \<langle>\<langle> qmsg) n)) TT"
+      hence "\<forall>i\<in>net_tree_ips n. ?inv (fst (initmissing (netgmap fst \<sigma>))) i"
+        by - (drule invariantD [OF proto],
+              simp only: aodv_openproc_par_qmsg.netglobalsimp
+                         fst_initmissing_netgmap_pair_fst)
+      thus "?inv (fst (initmissing (netgmap fst \<sigma>))) i"
+      proof (cases "i\<in>net_tree_ips n")
+        assume "i\<notin>net_tree_ips n"
+        from sr have "\<sigma> \<in> reachable (pnet (\<lambda>i. paodv i \<langle>\<langle> qmsg) n) TT" ..
+        hence "net_ips \<sigma> = net_tree_ips n" ..
+        with \<open>i\<notin>net_tree_ips n\<close> have "i\<notin>net_ips \<sigma>" by simp
+        hence "(fst (initmissing (netgmap fst \<sigma>))) i = aodv_init i"
+          by simp
+        thus ?thesis by simp
+      qed metis
+    qed
+  qed
+
+subsection \<open>Loop freedom of AODV\<close>
+
+theorem aodv_loop_freedom:
+  assumes "wf_net_tree n"
+  shows "closed (pnet (\<lambda>i. paodv i \<langle>\<langle> qmsg) n) \<TTurnstile> netglobal (\<lambda>\<sigma>. \<forall>dip. irrefl ((rt_graph \<sigma> dip)\<^sup>+))"
+  using assms by (rule aodv_openproc_par_qmsg.netglobal_weakenE
+                          [OF net_nhop_quality_increases inv_to_loop_freedom])
+
+end

formal/afp/AODV/variants/a_norreqid/A_Aodv_Message.thy

@@ -0,0 +1,75 @@
+(*  Title:       variants/a_norreqid/Aodv_Message.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+    Author:      Peter Höfner, NICTA
+*)
+
+section "AODV protocol messages"
+
+theory A_Aodv_Message
+imports A_Norreqid
+begin
+
+datatype msg =
+    Rreq nat ip sqn k ip sqn ip
+  | Rrep nat ip sqn ip ip
+  | Rerr "ip \<rightharpoonup> sqn" ip
+  | Newpkt data ip
+  | Pkt data ip ip
+
+instantiation msg :: msg
+begin
+  definition newpkt_def [simp]: "newpkt \<equiv> \<lambda>(d, dip). Newpkt d dip"
+  definition eq_newpkt_def: "eq_newpkt m \<equiv> case m of Newpkt d dip \<Rightarrow> True | _ \<Rightarrow> False"
+
+  instance by intro_classes (simp add: eq_newpkt_def)  
+end
+ 
+text \<open>The @{type msg} type models the different messages used within AODV.
+      The instantiation as a @{class msg} is a technicality due to the special
+      treatment of @{term newpkt} messages in the AWN SOS rules.
+      This use of classes allows a clean separation of the AWN-specific definitions
+      and these AODV-specific definitions.\<close>
+
+definition rreq :: "nat \<times> ip \<times> sqn \<times> k \<times> ip \<times> sqn \<times> ip \<Rightarrow> msg"
+  where "rreq \<equiv> \<lambda>(hops, dip, dsn, dsk, oip, osn, sip).
+                    Rreq hops dip dsn dsk oip osn sip"
+
+lemma rreq_simp [simp]:
+  "rreq(hops, dip, dsn, dsk, oip, osn, sip) =  Rreq hops dip dsn dsk oip osn sip"
+  unfolding rreq_def by simp
+
+definition rrep :: "nat \<times> ip \<times> sqn \<times> ip \<times> ip \<Rightarrow> msg"
+  where "rrep \<equiv> \<lambda>(hops, dip, dsn, oip, sip). Rrep hops dip dsn oip sip"
+
+lemma rrep_simp [simp]:
+  "rrep(hops, dip, dsn, oip, sip) = Rrep hops dip dsn oip sip"
+  unfolding rrep_def by simp
+
+definition rerr :: "(ip \<rightharpoonup> sqn) \<times> ip \<Rightarrow> msg"
+  where "rerr \<equiv> \<lambda>(dests, sip). Rerr dests sip"
+
+lemma rerr_simp [simp]:
+  "rerr(dests, sip) = Rerr dests sip"
+  unfolding rerr_def by simp
+
+lemma not_eq_newpkt_rreq [simp]: "\<not>eq_newpkt (Rreq hops dip dsn dsk oip osn sip)"
+  unfolding eq_newpkt_def by simp
+
+lemma not_eq_newpkt_rrep [simp]: "\<not>eq_newpkt (Rrep hops dip dsn oip sip)"
+  unfolding eq_newpkt_def by simp
+
+lemma not_eq_newpkt_rerr [simp]: "\<not>eq_newpkt (Rerr dests sip)"
+  unfolding eq_newpkt_def by simp
+
+lemma not_eq_newpkt_pkt [simp]: "\<not>eq_newpkt (Pkt d dip sip)"
+  unfolding eq_newpkt_def by simp
+
+definition pkt :: "data \<times> ip \<times> ip \<Rightarrow> msg"
+  where "pkt \<equiv> \<lambda>(d, dip, sip). Pkt d dip sip"
+
+lemma pkt_simp [simp]:
+  "pkt(d, dip, sip) = Pkt d dip sip"
+  unfolding pkt_def by simp
+
+end

formal/afp/AODV/variants/a_norreqid/A_Aodv_Predicates.thy

@@ -0,0 +1,137 @@
+(*  Title:       variants/a_norreqid/Aodv_Predicates.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+    Author:      Peter Höfner, NICTA
+*)
+
+section "Invariant assumptions and properties"
+
+theory A_Aodv_Predicates
+imports A_Aodv
+begin
+
+text \<open>Definitions for expression assumptions on incoming messages and properties of
+      outgoing messages.\<close>
+
+abbreviation not_Pkt :: "msg \<Rightarrow> bool"
+where "not_Pkt m \<equiv> case m of Pkt _ _ _ \<Rightarrow> False | _ \<Rightarrow> True"
+
+definition msg_sender :: "msg \<Rightarrow> ip"
+where "msg_sender m \<equiv> case m of Rreq _ _ _ _ _ _ ipc \<Rightarrow> ipc
+                              | Rrep _ _ _ _ ipc \<Rightarrow> ipc
+                              | Rerr _ ipc \<Rightarrow> ipc
+                              | Pkt _ _ ipc \<Rightarrow> ipc"
+
+lemma msg_sender_simps [simp]:
+  "\<And>hops dip dsn dsk oip osn sip.
+                          msg_sender (Rreq hops dip dsn dsk oip osn sip) = sip"
+  "\<And>hops dip dsn oip sip. msg_sender (Rrep hops dip dsn oip sip) = sip"
+  "\<And>dests sip.            msg_sender (Rerr dests sip) = sip"
+  "\<And>d dip sip.            msg_sender (Pkt d dip sip) = sip"
+  unfolding msg_sender_def by simp_all
+
+definition msg_zhops :: "msg \<Rightarrow> bool"
+where "msg_zhops m \<equiv> case m of
+                                 Rreq hopsc dipc _ _ oipc _ sipc \<Rightarrow> hopsc = 0 \<longrightarrow> oipc = sipc
+                               | Rrep hopsc dipc _ _ sipc \<Rightarrow> hopsc = 0 \<longrightarrow> dipc = sipc
+                               | _ \<Rightarrow> True"
+
+lemma msg_zhops_simps [simp]:
+  "\<And>hops dip dsn dsk oip osn sip.
+           msg_zhops (Rreq hops dip dsn dsk oip osn sip) = (hops = 0 \<longrightarrow> oip = sip)"
+  "\<And>hops dip dsn oip sip. msg_zhops (Rrep hops dip dsn oip sip) = (hops = 0 \<longrightarrow> dip = sip)"
+  "\<And>dests sip.            msg_zhops (Rerr dests sip)        = True"
+  "\<And>d dip.                msg_zhops (Newpkt d dip)          = True"
+  "\<And>d dip sip.            msg_zhops (Pkt d dip sip)         = True"
+  unfolding msg_zhops_def by simp_all
+
+definition rreq_rrep_sn :: "msg \<Rightarrow> bool"
+where "rreq_rrep_sn m \<equiv> case m of Rreq  _ _ _ _ _ osnc _ \<Rightarrow> osnc \<ge> 1
+                                | Rrep _ _ dsnc _ _ \<Rightarrow> dsnc \<ge> 1
+                                | _ \<Rightarrow> True"
+
+lemma rreq_rrep_sn_simps [simp]:
+  "\<And>hops dip dsn dsk oip osn sip.
+           rreq_rrep_sn (Rreq hops dip dsn dsk oip osn sip) = (osn \<ge> 1)"
+  "\<And>hops dip dsn oip sip. rreq_rrep_sn (Rrep hops dip dsn oip sip) = (dsn \<ge> 1)"
+  "\<And>dests sip.            rreq_rrep_sn (Rerr dests sip) = True"
+  "\<And>d dip.                rreq_rrep_sn (Newpkt d dip)   = True"
+  "\<And>d dip sip.            rreq_rrep_sn (Pkt d dip sip)  = True"
+  unfolding rreq_rrep_sn_def by simp_all
+
+definition rreq_rrep_fresh :: "rt \<Rightarrow> msg \<Rightarrow> bool"
+where "rreq_rrep_fresh crt m \<equiv> case m of Rreq hopsc  _ _ _ oipc osnc ipcc \<Rightarrow> (ipcc \<noteq> oipc \<longrightarrow>
+                                                oipc\<in>kD(crt) \<and> (sqn crt oipc > osnc
+                                                                \<or> (sqn crt oipc = osnc
+                                                                   \<and> the (dhops crt oipc) \<le> hopsc
+                                                                   \<and> the (flag crt oipc) = val)))
+                                | Rrep hopsc dipc dsnc _ ipcc \<Rightarrow> (ipcc \<noteq> dipc \<longrightarrow> 
+                                                                    dipc\<in>kD(crt)
+                                                                  \<and> sqn crt dipc = dsnc
+                                                                  \<and> the (dhops crt dipc) = hopsc
+                                                                  \<and> the (flag crt dipc) = val)
+                                | _ \<Rightarrow> True"
+
+lemma rreq_rrep_fresh [simp]:
+  "\<And>hops dip dsn dsk oip osn sip.
+           rreq_rrep_fresh crt (Rreq hops dip dsn dsk oip osn sip) =
+                               (sip \<noteq> oip \<longrightarrow> oip\<in>kD(crt)
+                                            \<and> (sqn crt oip > osn
+                                               \<or> (sqn crt oip = osn
+                                                  \<and> the (dhops crt oip) \<le> hops
+                                                  \<and> the (flag crt oip) = val)))"
+  "\<And>hops dip dsn oip sip. rreq_rrep_fresh crt (Rrep hops dip dsn oip sip) =
+                               (sip \<noteq> dip \<longrightarrow> dip\<in>kD(crt)
+                                              \<and> sqn crt dip = dsn
+                                              \<and> the (dhops crt dip) = hops
+                                              \<and> the (flag crt dip) = val)"
+  "\<And>dests sip.            rreq_rrep_fresh crt (Rerr dests sip) = True"
+  "\<And>d dip.                rreq_rrep_fresh crt (Newpkt d dip)   = True"
+  "\<And>d dip sip.            rreq_rrep_fresh crt (Pkt d dip sip)  = True"
+  unfolding rreq_rrep_fresh_def by simp_all
+
+definition rerr_invalid :: "rt \<Rightarrow> msg \<Rightarrow> bool"
+where "rerr_invalid crt m \<equiv> case m of Rerr destsc _ \<Rightarrow> (\<forall>ripc\<in>dom(destsc).
+                                            (ripc\<in>iD(crt) \<and> the (destsc ripc) = sqn crt ripc))
+                                | _ \<Rightarrow> True"
+
+lemma rerr_invalid [simp]:
+  "\<And>hops dip dsn dsk oip osn sip.
+                           rerr_invalid crt (Rreq hops dip dsn dsk oip osn sip) = True"
+  "\<And>hops dip dsn oip sip. rerr_invalid crt (Rrep hops dip dsn oip sip) = True"
+  "\<And>dests sip.            rerr_invalid crt (Rerr dests sip) = (\<forall>rip\<in>dom(dests).
+                                                 rip\<in>iD(crt) \<and> the (dests rip) = sqn crt rip)"
+  "\<And>d dip.                rerr_invalid crt (Newpkt d dip)   = True"
+  "\<And>d dip sip.            rerr_invalid crt (Pkt d dip sip)  = True"
+  unfolding rerr_invalid_def by simp_all
+
+definition
+  initmissing :: "(nat \<Rightarrow> state option) \<times> 'a \<Rightarrow> (nat \<Rightarrow> state) \<times> 'a"
+where
+  "initmissing \<sigma> = (\<lambda>i. case (fst \<sigma>) i of None \<Rightarrow> aodv_init i | Some s \<Rightarrow> s, snd \<sigma>)"
+
+lemma not_in_net_ips_fst_init_missing [simp]:
+  assumes "i \<notin> net_ips \<sigma>"
+    shows "fst (initmissing (netgmap fst \<sigma>)) i = aodv_init i"
+  using assms unfolding initmissing_def by simp
+
+lemma fst_initmissing_netgmap_pair_fst [simp]:
+  "fst (initmissing (netgmap (\<lambda>(p, q). (fst (id p), snd (id p), q)) s))
+                                               = fst (initmissing (netgmap fst s))"
+  unfolding initmissing_def by auto
+
+text \<open>We introduce a streamlined alternative to @{term initmissing} with @{term netgmap}
+        to simplify invariant statements and thus facilitate their comprehension and
+        presentation.\<close>
+
+lemma fst_initmissing_netgmap_default_aodv_init_netlift:
+  "fst (initmissing (netgmap fst s)) = default aodv_init (netlift fst s)"
+  unfolding initmissing_def default_def
+  by (simp add: fst_netgmap_netlift del: One_nat_def)
+
+definition
+  netglobal :: "((nat \<Rightarrow> state) \<Rightarrow> bool) \<Rightarrow> ((state \<times> 'b) \<times> 'c) net_state \<Rightarrow> bool"
+where
+  "netglobal P \<equiv> (\<lambda>s. P (default aodv_init (netlift fst s)))"
+
+end

formal/afp/AODV/variants/a_norreqid/A_Fresher.thy

@@ -0,0 +1,799 @@
+(*  Title:       variants/a_norreqid/Fresher.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "Quality relations between routes"
+
+theory A_Fresher
+imports A_Aodv_Data
+begin
+
+subsection "Net sequence numbers"
+
+subsubsection "On individual routes"
+
+definition
+  nsqn\<^sub>r :: "r \<Rightarrow> sqn"
+where
+  "nsqn\<^sub>r r \<equiv> if \<pi>\<^sub>4(r) = val \<or> \<pi>\<^sub>2(r) = 0 then \<pi>\<^sub>2(r) else (\<pi>\<^sub>2(r) - 1)"
+
+lemma nsqnr_def':
+  "nsqn\<^sub>r r = (if \<pi>\<^sub>4(r) = inv then \<pi>\<^sub>2(r) - 1 else \<pi>\<^sub>2(r))"
+  unfolding nsqn\<^sub>r_def by simp
+
+lemma nsqn\<^sub>r_zero [simp]:
+  "\<And>dsn dsk flag hops nhip pre. nsqn\<^sub>r (0, dsk, flag, hops, nhip, pre) = 0"
+  unfolding nsqn\<^sub>r_def by clarsimp
+
+lemma nsqn\<^sub>r_val [simp]:
+  "\<And>dsn dsk hops nhip pre. nsqn\<^sub>r (dsn, dsk, val, hops, nhip, pre) = dsn"
+  unfolding nsqn\<^sub>r_def by clarsimp
+
+lemma nsqn\<^sub>r_inv [simp]:
+  "\<And>dsn dsk hops nhip pre. nsqn\<^sub>r (dsn, dsk, inv, hops, nhip, pre) = dsn - 1"
+  unfolding nsqn\<^sub>r_def by clarsimp
+
+lemma nsqn\<^sub>r_lte_dsn [simp]:
+  "\<And>dsn dsk flag hops nhip pre. nsqn\<^sub>r (dsn, dsk, flag, hops, nhip, pre) \<le> dsn"
+  unfolding nsqn\<^sub>r_def by clarsimp
+
+subsubsection "On routes in routing tables"
+
+definition
+  nsqn :: "rt \<Rightarrow> ip \<Rightarrow> sqn"
+where
+  "nsqn \<equiv> \<lambda>rt dip. case \<sigma>\<^bsub>route\<^esub>(rt, dip) of None \<Rightarrow> 0 | Some r \<Rightarrow> nsqn\<^sub>r(r)"
+
+lemma nsqn_sqn_def:
+  "\<And>rt dip. nsqn rt dip = (if flag rt dip = Some val \<or> sqn rt dip = 0
+                            then sqn rt dip else sqn rt dip - 1)"
+  unfolding nsqn_def sqn_def by (clarsimp split: option.split)
+
+lemma not_in_kD_nsqn [simp]:
+  assumes "dip \<notin> kD(rt)"
+  shows "nsqn rt dip = 0"
+  using assms unfolding nsqn_def by simp
+
+lemma kD_nsqn:
+  assumes "dip \<in> kD(rt)"
+    shows "nsqn rt dip = nsqn\<^sub>r(the (\<sigma>\<^bsub>route\<^esub>(rt, dip)))"
+  using assms [THEN kD_Some] unfolding nsqn_def by clarsimp
+
+lemma nsqnr_r_flag_pred [simp, intro]:
+    fixes dsn dsk flag hops nhip pre
+  assumes "P (nsqn\<^sub>r (dsn, dsk, val, hops, nhip, pre))"
+      and "P (nsqn\<^sub>r (dsn, dsk, inv, hops, nhip, pre))"
+    shows "P (nsqn\<^sub>r (dsn, dsk, flag, hops, nhip, pre))"
+  using assms by (cases flag) auto
+
+lemma nsqn\<^sub>r_addpreRT_inv [simp]:
+  "\<And>rt dip npre dip'. dip \<in> kD(rt) \<Longrightarrow>
+   nsqn\<^sub>r (the (the (addpreRT rt dip npre) dip')) = nsqn\<^sub>r (the (rt dip'))"
+  unfolding addpreRT_def nsqn\<^sub>r_def
+  by (frule kD_Some) (clarsimp split: option.split)
+
+lemma sqn_nsqn:
+  "\<And>rt dip. sqn rt dip - 1 \<le> nsqn rt dip"
+  unfolding sqn_def nsqn_def by (clarsimp split: option.split)
+
+lemma nsqn_sqn: "nsqn rt dip \<le> sqn rt dip"
+  unfolding sqn_def nsqn_def by (cases "rt dip") auto
+
+lemma val_nsqn_sqn [elim, simp]:
+  assumes "ip\<in>kD(rt)"
+      and "the (flag rt ip) = val"
+    shows "nsqn rt ip = sqn rt ip"
+  using assms unfolding nsqn_sqn_def by auto
+
+lemma vD_nsqn_sqn [elim, simp]:
+  assumes "ip\<in>vD(rt)"
+    shows "nsqn rt ip = sqn rt ip"
+  proof -
+    from \<open>ip\<in>vD(rt)\<close> have "ip\<in>kD(rt)"
+                      and "the (flag rt ip) = val" by auto
+    thus ?thesis ..
+  qed
+
+lemma inv_nsqn_sqn [elim, simp]:
+  assumes "ip\<in>kD(rt)"
+      and "the (flag rt ip) = inv"
+    shows "nsqn rt ip = sqn rt ip - 1"
+  using assms unfolding nsqn_sqn_def by auto
+
+lemma iD_nsqn_sqn [elim, simp]:
+  assumes "ip\<in>iD(rt)"
+    shows "nsqn rt ip = sqn rt ip - 1"
+  proof -
+    from \<open>ip\<in>iD(rt)\<close> have "ip\<in>kD(rt)"
+                      and "the (flag rt ip) = inv" by auto
+    thus ?thesis ..
+  qed
+
+lemma nsqn_update_changed_kno_val [simp]: "\<And>rt ip dsn dsk hops nhip.
+  rt \<noteq> update rt ip (dsn, kno, val, hops, nhip, {})
+   \<Longrightarrow> nsqn (update rt ip (dsn, kno, val, hops, nhip, {})) ip = dsn"
+  unfolding nsqn\<^sub>r_def update_def
+  by (clarsimp simp: kD_nsqn split: option.split_asm option.split if_split_asm)
+     (metis fun_upd_triv)
+
+lemma nsqn_addpreRT_inv [simp]:
+  "\<And>rt dip npre dip'. dip \<in> kD(rt) \<Longrightarrow>
+   nsqn (the (addpreRT rt dip npre)) dip' = nsqn rt dip'"
+  unfolding addpreRT_def nsqn_def nsqn\<^sub>r_def
+  by (frule kD_Some) (clarsimp split: option.split)
+
+lemma nsqn_update_other [simp]:
+    fixes dsn dsk flag hops dip nhip pre rt ip
+  assumes "dip \<noteq> ip"
+    shows "nsqn (update rt ip (dsn, dsk, flag, hops, nhip, pre)) dip = nsqn rt dip"
+    using assms unfolding nsqn_def
+    by (clarsimp split: option.split)
+
+lemma nsqn_invalidate_eq:
+  assumes "dip \<in> kD(rt)"
+      and "dests dip = Some rsn"
+    shows "nsqn (invalidate rt dests) dip = rsn - 1"
+  using assms
+  proof -
+    from assms obtain dsk hops nhip pre
+      where "invalidate rt dests dip = Some (rsn, dsk, inv, hops, nhip, pre)"
+      unfolding invalidate_def
+      by auto
+    moreover from \<open>dip \<in> kD(rt)\<close> have "dip \<in> kD(invalidate rt dests)" by simp
+    ultimately show ?thesis
+      using \<open>dests dip = Some rsn\<close> by simp
+  qed
+
+lemma nsqn_invalidate_other [simp]:
+  assumes "dip\<in>kD(rt)"
+      and "dip\<notin>dom dests"
+    shows "nsqn (invalidate rt dests) dip = nsqn rt dip"
+  using assms by (clarsimp simp add: kD_nsqn)
+
+subsection "Comparing routes "
+
+definition
+  fresher :: "r \<Rightarrow> r \<Rightarrow> bool" ("(_/ \<sqsubseteq> _)"  [51, 51] 50)
+where
+  "fresher r r' \<equiv> ((nsqn\<^sub>r r < nsqn\<^sub>r r') \<or> (nsqn\<^sub>r r  = nsqn\<^sub>r r' \<and> \<pi>\<^sub>5(r) \<ge> \<pi>\<^sub>5(r')))"
+
+lemma fresherI1 [intro]:
+  assumes "nsqn\<^sub>r r < nsqn\<^sub>r r'"
+    shows "r \<sqsubseteq> r'"
+  unfolding fresher_def using assms by simp
+
+lemma fresherI2 [intro]:
+  assumes "nsqn\<^sub>r r = nsqn\<^sub>r r'"
+      and "\<pi>\<^sub>5(r) \<ge> \<pi>\<^sub>5(r')"
+    shows "r \<sqsubseteq> r'"
+  unfolding fresher_def using assms by simp
+
+lemma fresherI [intro]:
+  assumes "(nsqn\<^sub>r r < nsqn\<^sub>r r') \<or> (nsqn\<^sub>r r  = nsqn\<^sub>r r' \<and> \<pi>\<^sub>5(r) \<ge> \<pi>\<^sub>5(r'))"
+    shows "r \<sqsubseteq> r'"
+  unfolding fresher_def using assms .
+
+lemma fresherE [elim]:
+  assumes "r \<sqsubseteq> r'"
+      and "nsqn\<^sub>r r < nsqn\<^sub>r r' \<Longrightarrow> P r r'"
+      and "nsqn\<^sub>r r  = nsqn\<^sub>r r' \<and> \<pi>\<^sub>5(r) \<ge> \<pi>\<^sub>5(r') \<Longrightarrow> P r r'"
+    shows "P r r'"
+  using assms unfolding fresher_def by auto
+
+lemma fresher_refl [simp]: "r \<sqsubseteq> r"
+  unfolding fresher_def by simp
+
+lemma fresher_trans [elim, trans]:
+  "\<lbrakk> x \<sqsubseteq> y; y \<sqsubseteq> z \<rbrakk> \<Longrightarrow> x \<sqsubseteq> z"
+  unfolding fresher_def by auto
+
+lemma not_fresher_trans [elim, trans]:
+  "\<lbrakk> \<not>(x \<sqsubseteq> y); \<not>(z \<sqsubseteq> x) \<rbrakk> \<Longrightarrow> \<not>(z \<sqsubseteq> y)"
+  unfolding fresher_def by auto
+
+lemma fresher_dsn_flag_hops_const [simp]:
+  fixes dsn dsk dsk' flag hops nhip nhip' pre pre'
+  shows "(dsn, dsk, flag, hops, nhip, pre) \<sqsubseteq> (dsn, dsk', flag, hops, nhip', pre')"
+  unfolding fresher_def by (cases flag) simp_all
+
+lemma addpre_fresher [simp]: "\<And>r npre. r \<sqsubseteq> (addpre r npre)"
+  by clarsimp
+
+subsection "Comparing routing tables "
+
+definition
+  rt_fresher :: "ip \<Rightarrow> rt \<Rightarrow> rt \<Rightarrow> bool"
+where
+  "rt_fresher \<equiv> \<lambda>dip rt rt'. (the (\<sigma>\<^bsub>route\<^esub>(rt, dip))) \<sqsubseteq> (the (\<sigma>\<^bsub>route\<^esub>(rt', dip)))"
+
+abbreviation
+   rt_fresher_syn :: "rt \<Rightarrow> ip \<Rightarrow> rt \<Rightarrow> bool" ("(_/ \<sqsubseteq>\<^bsub>_\<^esub> _)"  [51, 999, 51] 50)
+where
+  "rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2 \<equiv> rt_fresher i rt1 rt2"
+
+lemma rt_fresher_def':
+  "(rt\<^sub>1 \<sqsubseteq>\<^bsub>i\<^esub> rt\<^sub>2) = (nsqn\<^sub>r (the (rt\<^sub>1 i)) < nsqn\<^sub>r (the (rt\<^sub>2 i)) \<or>
+                     nsqn\<^sub>r (the (rt\<^sub>1 i)) = nsqn\<^sub>r (the (rt\<^sub>2 i)) \<and> \<pi>\<^sub>5 (the (rt\<^sub>2 i)) \<le> \<pi>\<^sub>5 (the (rt\<^sub>1 i)))"
+  unfolding rt_fresher_def fresher_def by (rule refl)
+
+lemma single_rt_fresher [intro]:
+  assumes "the (rt1 ip) \<sqsubseteq> the (rt2 ip)"
+    shows "rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2"
+  using assms unfolding rt_fresher_def .
+
+lemma rt_fresher_single [intro]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2"
+    shows "the (rt1 ip) \<sqsubseteq> the (rt2 ip)"
+  using assms unfolding rt_fresher_def .
+
+lemma rt_fresher_def2:
+  assumes "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+    shows "(rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2) = (nsqn rt1 dip < nsqn rt2 dip
+                               \<or> (nsqn rt1 dip = nsqn rt2 dip
+                                   \<and> the (dhops rt1 dip) \<ge> the (dhops rt2 dip)))"
+  using assms unfolding rt_fresher_def fresher_def by (simp add: kD_nsqn proj5_eq_dhops)
+
+lemma rt_fresherI1 [intro]:
+  assumes "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+      and "nsqn rt1 dip < nsqn rt2 dip"
+    shows "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+  unfolding rt_fresher_def2 [OF assms(1-2)] using assms(3) by simp
+
+lemma rt_fresherI2 [intro]:
+  assumes "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+      and "nsqn rt1 dip = nsqn rt2 dip"
+      and "the (dhops rt1 dip) \<ge> the (dhops rt2 dip)"
+    shows "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+  unfolding rt_fresher_def2 [OF assms(1-2)] using assms(3-4) by simp
+
+lemma rt_fresherE [elim]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+      and "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+      and "\<lbrakk> nsqn rt1 dip < nsqn rt2 dip \<rbrakk> \<Longrightarrow> P rt1 rt2 dip"
+      and "\<lbrakk> nsqn rt1 dip = nsqn rt2 dip;
+             the (dhops rt1 dip) \<ge> the (dhops rt2 dip) \<rbrakk> \<Longrightarrow> P rt1 rt2 dip"
+    shows "P rt1 rt2 dip"
+  using assms(1) unfolding rt_fresher_def2 [OF assms(2-3)]
+  using assms(4-5) by auto
+
+lemma rt_fresher_refl [simp]: "rt \<sqsubseteq>\<^bsub>dip\<^esub> rt"
+  unfolding rt_fresher_def by simp
+
+lemma rt_fresher_trans [elim, trans]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+      and "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt3"
+    shows "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt3"
+  using assms unfolding rt_fresher_def by auto
+
+lemma rt_fresher_if_Some [intro!]:
+  assumes "the (rt dip) \<sqsubseteq> r"
+    shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> (\<lambda>ip. if ip = dip then Some r else rt ip)"
+  using assms unfolding rt_fresher_def by simp
+
+definition rt_fresh_as :: "ip \<Rightarrow> rt \<Rightarrow> rt \<Rightarrow> bool"
+where
+  "rt_fresh_as \<equiv> \<lambda>dip rt1 rt2. (rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2) \<and> (rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)"
+
+abbreviation
+   rt_fresh_as_syn :: "rt \<Rightarrow> ip \<Rightarrow> rt \<Rightarrow> bool" ("(_/ \<approx>\<^bsub>_\<^esub> _)"  [51, 999, 51] 50)
+where
+  "rt1 \<approx>\<^bsub>i\<^esub> rt2 \<equiv> rt_fresh_as i rt1 rt2"
+
+lemma rt_fresh_as_refl [simp]: "\<And>rt dip. rt \<approx>\<^bsub>dip\<^esub> rt"
+  unfolding rt_fresh_as_def by simp
+
+lemma rt_fresh_as_trans [simp, intro, trans]:
+  "\<And>rt1 rt2 rt3 dip. \<lbrakk> rt1 \<approx>\<^bsub>dip\<^esub> rt2; rt2 \<approx>\<^bsub>dip\<^esub> rt3 \<rbrakk> \<Longrightarrow> rt1 \<approx>\<^bsub>dip\<^esub> rt3"
+  unfolding rt_fresh_as_def rt_fresher_def
+  by (metis (mono_tags) fresher_trans)
+
+lemma rt_fresh_asI [intro!]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+      and "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1"
+    shows "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+  using assms unfolding rt_fresh_as_def by simp
+
+lemma rt_fresh_as_fresherI [intro]:
+  assumes "dip\<in>kD(rt1)"
+      and "dip\<in>kD(rt2)"
+      and "the (rt1 dip) \<sqsubseteq> the (rt2 dip)"
+      and "the (rt2 dip) \<sqsubseteq> the (rt1 dip)"
+    shows "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+  using assms unfolding rt_fresh_as_def
+  by (clarsimp dest!: single_rt_fresher)
+
+lemma nsqn_rt_fresh_asI:
+  assumes "dip \<in> kD(rt)"
+      and "dip \<in> kD(rt')"
+      and "nsqn rt dip = nsqn rt' dip"
+      and "\<pi>\<^sub>5(the (rt dip)) = \<pi>\<^sub>5(the (rt' dip))"
+    shows "rt \<approx>\<^bsub>dip\<^esub> rt'"
+  proof
+    from assms(1-2,4) have dhops': "the (dhops rt' dip) \<le> the (dhops rt dip)"
+      by (simp add: proj5_eq_dhops)
+    with assms(1-3) show "rt \<sqsubseteq>\<^bsub>dip\<^esub> rt'"
+      by (rule rt_fresherI2)
+  next
+    from assms(1-2,4) have dhops: "the (dhops rt dip) \<le> the (dhops rt' dip)"
+      by (simp add: proj5_eq_dhops)
+    with assms(2,1) assms(3) [symmetric] show "rt' \<sqsubseteq>\<^bsub>dip\<^esub> rt"
+      by (rule rt_fresherI2)
+  qed
+
+lemma rt_fresh_asE [elim]:
+  assumes "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+      and "\<lbrakk> rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2; rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1 \<rbrakk> \<Longrightarrow> P rt1 rt2 dip"
+    shows "P rt1 rt2 dip"
+  using assms unfolding rt_fresh_as_def by simp
+
+lemma rt_fresh_asD1 [dest]:
+  assumes "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    shows "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+  using assms unfolding rt_fresh_as_def by simp
+
+lemma rt_fresh_asD2 [dest]:
+  assumes "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    shows "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1"
+  using assms unfolding rt_fresh_as_def by simp
+
+lemma rt_fresh_as_sym:
+  assumes "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    shows "rt2 \<approx>\<^bsub>dip\<^esub> rt1"
+  using assms unfolding rt_fresh_as_def by simp
+
+lemma not_rt_fresh_asI1 [intro]:
+  assumes "\<not> (rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2)"
+    shows "\<not> (rt1 \<approx>\<^bsub>dip\<^esub> rt2)"
+  proof
+    assume "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    hence "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2" ..
+    with \<open>\<not> (rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2)\<close> show False ..
+  qed
+
+lemma not_rt_fresh_asI2 [intro]:
+  assumes "\<not> (rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)"
+    shows "\<not> (rt1 \<approx>\<^bsub>dip\<^esub> rt2)"
+  proof
+    assume "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    hence "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1" ..
+    with \<open>\<not> (rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)\<close> show False ..
+  qed
+
+lemma not_single_rt_fresher [elim]:
+  assumes "\<not>(the (rt1 ip) \<sqsubseteq> the (rt2 ip))"
+    shows "\<not>(rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2)"
+  proof
+    assume "rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2"
+    hence "the (rt1 ip) \<sqsubseteq> the (rt2 ip)" ..
+    with \<open>\<not>(the (rt1 ip) \<sqsubseteq> the (rt2 ip))\<close> show False ..
+  qed
+
+lemmas not_single_rt_fresh_asI1 [intro] =  not_rt_fresh_asI1 [OF not_single_rt_fresher]
+lemmas not_single_rt_fresh_asI2 [intro] =  not_rt_fresh_asI2 [OF not_single_rt_fresher]
+
+lemma not_rt_fresher_single [elim]:
+  assumes "\<not>(rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2)"
+    shows "\<not>(the (rt1 ip) \<sqsubseteq> the (rt2 ip))"
+  proof
+    assume "the (rt1 ip) \<sqsubseteq> the (rt2 ip)"
+    hence "rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2" ..
+    with \<open>\<not>(rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2)\<close> show False ..
+  qed
+
+lemma rt_fresh_as_nsqnr:
+  assumes "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+      and "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    shows "nsqn\<^sub>r (the (rt2 dip)) = nsqn\<^sub>r (the (rt1 dip))"
+  using assms(3) unfolding rt_fresh_as_def
+  by (auto simp: rt_fresher_def2 [OF \<open>dip \<in> kD(rt1)\<close> \<open>dip \<in> kD(rt2)\<close>]
+                 rt_fresher_def2 [OF \<open>dip \<in> kD(rt2)\<close> \<open>dip \<in> kD(rt1)\<close>]
+                 kD_nsqn [OF \<open>dip \<in> kD(rt1)\<close>]
+                 kD_nsqn [OF \<open>dip \<in> kD(rt2)\<close>])
+
+lemma rt_fresher_mapupd [intro!]:
+  assumes "dip\<in>kD(rt)"
+      and "the (rt dip) \<sqsubseteq> r"
+    shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> rt(dip \<mapsto> r)"
+  using assms unfolding rt_fresher_def by simp
+
+lemma rt_fresher_map_update_other [intro!]:
+  assumes "dip\<in>kD(rt)"
+      and "dip \<noteq> ip"
+    shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> rt(ip \<mapsto> r)"
+  using assms unfolding rt_fresher_def by simp
+
+lemma rt_fresher_update_other [simp]:
+  assumes inkD: "dip\<in>kD(rt)"
+     and "dip \<noteq> ip"
+   shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> update rt ip r"
+   using assms unfolding update_def
+   by (clarsimp split: option.split) (fastforce)
+
+theorem rt_fresher_update [simp]:
+  assumes "dip\<in>kD(rt)"
+      and "the (dhops rt dip) \<ge> 1"
+      and "update_arg_wf r"
+   shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> update rt ip r"
+  proof (cases "dip = ip")
+    assume "dip \<noteq> ip" with \<open>dip\<in>kD(rt)\<close> show ?thesis
+      by (rule rt_fresher_update_other)
+  next
+    assume "dip = ip"
+
+    from \<open>dip\<in>kD(rt)\<close> obtain dsn\<^sub>n dsk\<^sub>n f\<^sub>n hops\<^sub>n nhip\<^sub>n pre\<^sub>n
+      where rtn [simp]: "the (rt dip) = (dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)"
+      by (metis prod_cases6)
+    with \<open>the (dhops rt dip) \<ge> 1\<close> and \<open>dip\<in>kD(rt)\<close> have "hops\<^sub>n \<ge> 1"
+      by (metis proj5_eq_dhops projs(4))
+    from \<open>dip\<in>kD(rt)\<close> rtn have [simp]: "sqn rt dip = dsn\<^sub>n"
+                           and [simp]: "the (dhops rt dip) = hops\<^sub>n"
+                           and [simp]: "the (flag rt dip) = f\<^sub>n"
+      by (simp add: sqn_def proj5_eq_dhops [symmetric]
+                            proj4_eq_flag [symmetric])+
+
+    from \<open>update_arg_wf r\<close> have "(dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+                                  \<sqsubseteq> the ((update rt dip r) dip)"
+      proof (rule wf_r_cases)
+        fix nhip pre
+        from \<open>hops\<^sub>n \<ge> 1\<close> have "\<And>pre'. (dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+                                        \<sqsubseteq> (dsn\<^sub>n, unk, val, Suc 0, nhip, pre')"
+          unfolding fresher_def sqn_def by (cases f\<^sub>n) auto
+        thus "(dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+               \<sqsubseteq> the (update rt dip (0, unk, val, Suc 0, nhip, pre) dip)"
+          using \<open>dip\<in>kD(rt)\<close> by - (rule update_cases_kD, simp_all)
+      next
+        fix dsn :: sqn and hops nhip pre
+        assume "0 < dsn"
+        show "(dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+               \<sqsubseteq> the (update rt dip (dsn, kno, val, hops, nhip, pre) dip)"
+        proof (rule update_cases_kD [OF _ \<open>dip\<in>kD(rt)\<close>], simp_all add: \<open>0 < dsn\<close>)
+          assume "dsn\<^sub>n < dsn"
+            thus "(dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+                   \<sqsubseteq> (dsn, kno, val, hops, nhip, pre \<union> pre\<^sub>n)"
+              unfolding fresher_def by auto
+        next
+          assume "dsn\<^sub>n = dsn"
+             and "hops < hops\<^sub>n"
+            thus "(dsn, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+                   \<sqsubseteq> (dsn, kno, val, hops, nhip, pre \<union> pre\<^sub>n)"
+              unfolding fresher_def nsqn\<^sub>r_def by simp
+        next
+          assume "dsn\<^sub>n = dsn"
+            with \<open>0 < dsn\<close>
+            show "(dsn, dsk\<^sub>n, inv, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+                   \<sqsubseteq> (dsn, kno, val, hops, nhip, pre \<union> pre\<^sub>n)"
+              unfolding fresher_def by simp
+        qed
+      qed
+    hence "rt \<sqsubseteq>\<^bsub>dip\<^esub> update rt dip r"
+      by - (rule single_rt_fresher, simp)
+    with \<open>dip = ip\<close> show ?thesis by simp
+  qed
+
+theorem rt_fresher_invalidate [simp]:
+  assumes "dip\<in>kD(rt)"
+      and indests: "\<forall>rip\<in>dom(dests). rip\<in>vD(rt) \<and> sqn rt rip < the (dests rip)"
+    shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> invalidate rt dests"
+  proof (cases "dip\<in>dom(dests)")
+    assume "dip\<notin>dom(dests)"
+      thus ?thesis using \<open>dip\<in>kD(rt)\<close>
+      by - (rule single_rt_fresher, simp)
+  next
+    assume "dip\<in>dom(dests)"
+    moreover with indests have "dip\<in>vD(rt)"
+                           and "sqn rt dip < the (dests dip)"
+      by auto
+    ultimately show ?thesis
+      unfolding invalidate_def sqn_def
+      by - (rule single_rt_fresher, auto simp: fresher_def)
+  qed
+
+lemma nsqn\<^sub>r_invalidate [simp]:
+  assumes "dip\<in>kD(rt)"
+      and "dip\<in>dom(dests)"
+    shows "nsqn\<^sub>r (the (invalidate rt dests dip)) = the (dests dip) - 1"
+  using assms unfolding invalidate_def by auto
+
+lemma rt_fresh_as_inc_invalidate [simp]:
+  assumes "dip\<in>kD(rt)"
+      and "\<forall>rip\<in>dom(dests). rip\<in>vD(rt) \<and> the (dests rip) = inc (sqn rt rip)"
+    shows "rt \<approx>\<^bsub>dip\<^esub> invalidate rt dests"
+  proof (cases "dip\<in>dom(dests)")
+    assume "dip\<notin>dom(dests)"
+    with \<open>dip\<in>kD(rt)\<close> have "dip\<in>kD(invalidate rt dests)"
+      by simp
+    with \<open>dip\<in>kD(rt)\<close> show ?thesis
+      by rule (simp_all add: \<open>dip\<notin>dom(dests)\<close>)
+  next
+    assume "dip\<in>dom(dests)"
+    with assms(2) have "dip\<in>vD(rt)"
+                  and "the (dests dip) = inc (sqn rt dip)" by auto
+    from \<open>dip\<in>vD(rt)\<close> have "dip\<in>kD(rt)" by simp
+    moreover then have "dip\<in>kD(invalidate rt dests)" by simp
+    ultimately show ?thesis
+    proof (rule nsqn_rt_fresh_asI)
+      from \<open>dip\<in>vD(rt)\<close> have "nsqn rt dip = sqn rt dip" by simp
+      also have "sqn rt dip = nsqn\<^sub>r (the (invalidate rt dests dip))"
+      proof -
+        from \<open>dip\<in>kD(rt)\<close> have "nsqn\<^sub>r (the (invalidate rt dests dip)) = the (dests dip) - 1"
+          using \<open>dip\<in>dom(dests)\<close> by (rule nsqn\<^sub>r_invalidate)
+        with \<open>the (dests dip) = inc (sqn rt dip)\<close>
+          show "sqn rt dip = nsqn\<^sub>r (the (invalidate rt dests dip))" by simp
+      qed
+      also from \<open>dip\<in>kD(invalidate rt dests)\<close>
+        have "nsqn\<^sub>r (the (invalidate rt dests dip)) = nsqn (invalidate rt dests) dip"
+          by (simp add: kD_nsqn)
+      finally show "nsqn rt dip = nsqn (invalidate rt dests) dip" .
+    qed simp
+  qed
+
+lemmas rt_fresher_inc_invalidate [simp] = rt_fresh_as_inc_invalidate [THEN rt_fresh_asD1]
+
+lemma rt_fresh_as_addpreRT [simp]:
+  assumes "ip\<in>kD(rt)"
+    shows "rt \<approx>\<^bsub>dip\<^esub> the (addpreRT rt ip npre)"
+  using assms [THEN kD_Some] by (auto simp: addpreRT_def)
+
+lemmas rt_fresher_addpreRT [simp] = rt_fresh_as_addpreRT [THEN rt_fresh_asD1]
+
+subsection "Strictly comparing routing tables "
+
+definition rt_strictly_fresher :: "ip \<Rightarrow> rt \<Rightarrow> rt \<Rightarrow> bool"
+where
+  "rt_strictly_fresher \<equiv> \<lambda>dip rt1 rt2. (rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2) \<and> \<not>(rt1 \<approx>\<^bsub>dip\<^esub> rt2)"
+
+abbreviation
+   rt_strictly_fresher_syn :: "rt \<Rightarrow> ip \<Rightarrow> rt \<Rightarrow> bool" ("(_/ \<sqsubset>\<^bsub>_\<^esub> _)"  [51, 999, 51] 50)
+where
+  "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2 \<equiv> rt_strictly_fresher i rt1 rt2"
+
+lemma rt_strictly_fresher_def'':
+  "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2 = ((rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2) \<and> \<not>(rt2 \<sqsubseteq>\<^bsub>i\<^esub> rt1))"
+  unfolding rt_strictly_fresher_def rt_fresh_as_def by auto
+
+lemma rt_strictly_fresherI' [intro]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2"
+      and "\<not>(rt2 \<sqsubseteq>\<^bsub>i\<^esub> rt1)"
+    shows "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2"
+  using assms unfolding rt_strictly_fresher_def'' by simp
+
+lemma rt_strictly_fresherE' [elim]:
+  assumes "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2"
+      and "\<lbrakk> rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2; \<not>(rt2 \<sqsubseteq>\<^bsub>i\<^esub> rt1) \<rbrakk> \<Longrightarrow> P rt1 rt2 i"
+    shows "P rt1 rt2 i"
+  using assms unfolding rt_strictly_fresher_def'' by simp
+
+lemma rt_strictly_fresherI [intro]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2"
+      and "\<not>(rt1 \<approx>\<^bsub>i\<^esub> rt2)"
+    shows "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2"
+  unfolding rt_strictly_fresher_def using assms ..
+
+lemmas rt_strictly_fresher_singleI [elim] = rt_strictly_fresherI [OF single_rt_fresher]
+
+lemma rt_strictly_fresherE [elim]:
+  assumes "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2"
+      and "\<lbrakk> rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2; \<not>(rt1 \<approx>\<^bsub>i\<^esub> rt2) \<rbrakk> \<Longrightarrow> P rt1 rt2 i"
+    shows "P rt1 rt2 i"
+  using assms(1) unfolding rt_strictly_fresher_def
+  by rule (erule(1) assms(2))
+
+lemma rt_strictly_fresher_def':
+  "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2 =
+     (nsqn\<^sub>r (the (rt1 i)) < nsqn\<^sub>r (the (rt2 i))
+       \<or> (nsqn\<^sub>r (the (rt1 i)) = nsqn\<^sub>r (the (rt2 i)) \<and> \<pi>\<^sub>5(the (rt1 i)) > \<pi>\<^sub>5(the (rt2 i))))"
+  unfolding rt_strictly_fresher_def'' rt_fresher_def fresher_def by auto
+
+lemma rt_strictly_fresher_fresherD [dest]:
+  assumes "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2"
+    shows "the (rt1 dip) \<sqsubseteq> the (rt2 dip)"
+  using assms unfolding rt_strictly_fresher_def rt_fresher_def by auto
+
+lemma rt_strictly_fresher_not_fresh_asD [dest]:
+  assumes "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2"
+    shows "\<not> rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+  using assms unfolding rt_strictly_fresher_def by auto
+
+lemma rt_strictly_fresher_trans [elim, trans]:
+  assumes "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2"
+      and "rt2 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+    shows "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+  using assms proof -
+    from \<open>rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2\<close> obtain "the (rt1 dip) \<sqsubseteq> the (rt2 dip)" by auto
+    also from \<open>rt2 \<sqsubset>\<^bsub>dip\<^esub> rt3\<close> obtain "the (rt2 dip) \<sqsubseteq> the (rt3 dip)" by auto
+    finally have "the (rt1 dip) \<sqsubseteq> the (rt3 dip)" .
+
+    moreover have "\<not> (rt1 \<approx>\<^bsub>dip\<^esub> rt3)"
+    proof -    
+      from \<open>rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2\<close> obtain "\<not>(the (rt2 dip) \<sqsubseteq> the (rt1 dip))" by auto
+      also from \<open>rt2 \<sqsubset>\<^bsub>dip\<^esub> rt3\<close> obtain "\<not>(the (rt3 dip) \<sqsubseteq> the (rt2 dip))" by auto
+      finally have "\<not>(the (rt3 dip) \<sqsubseteq> the (rt1 dip))" .
+      thus ?thesis ..
+    qed
+    ultimately show "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3" ..
+ qed
+
+lemma rt_strictly_fresher_irefl [simp]: "\<not> (rt \<sqsubset>\<^bsub>dip\<^esub> rt)"
+  unfolding rt_strictly_fresher_def
+  by clarsimp
+
+lemma rt_fresher_trans_rt_strictly_fresher [elim, trans]:
+  assumes "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2"
+      and "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt3"
+    shows "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+  proof -
+    from \<open>rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2\<close> have "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+                           and "\<not>(rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)"
+      unfolding rt_strictly_fresher_def'' by auto
+    from this(1) and \<open>rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt3\<close> have "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt3" ..
+
+    moreover from \<open>\<not>(rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)\<close> have "\<not>(rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)"
+      proof (rule contrapos_nn)
+        assume "rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt1"
+        with \<open>rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt3\<close> show "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1" ..
+      qed
+
+    ultimately show "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+      unfolding rt_strictly_fresher_def'' by auto
+  qed
+
+lemma rt_fresher_trans_rt_strictly_fresher' [elim, trans]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+      and "rt2 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+    shows "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+  proof -
+    from \<open>rt2 \<sqsubset>\<^bsub>dip\<^esub> rt3\<close> have "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt3"
+                           and "\<not>(rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt2)"
+      unfolding rt_strictly_fresher_def'' by auto
+    from \<open>rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2\<close> and this(1) have "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt3" ..
+
+    moreover from \<open>\<not>(rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt2)\<close> have "\<not>(rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)"
+      proof (rule contrapos_nn)
+        assume "rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt1"
+        thus "rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt2" using \<open>rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2\<close> ..
+      qed
+
+    ultimately show "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+      unfolding rt_strictly_fresher_def'' by auto
+  qed
+
+lemma rt_fresher_imp_nsqn_le:
+  assumes "rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2"
+      and "ip \<in> kD rt1"
+      and "ip \<in> kD rt2"
+    shows "nsqn rt1 ip \<le> nsqn rt2 ip"
+  using assms(1)
+  by (auto simp add: rt_fresher_def2 [OF assms(2-3)])
+
+lemma rt_strictly_fresher_ltI [intro]:
+  assumes "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+      and "nsqn rt1 dip < nsqn rt2 dip"
+    shows "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2"
+  proof
+    from assms show "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2" ..
+  next
+    show "\<not>(rt1 \<approx>\<^bsub>dip\<^esub> rt2)"
+    proof
+      assume "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+      hence "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1" ..
+      hence "nsqn rt2 dip \<le> nsqn rt1 dip"
+        using \<open>dip \<in> kD(rt2)\<close> \<open>dip \<in> kD(rt1)\<close>
+        by (rule rt_fresher_imp_nsqn_le)
+      with \<open>nsqn rt1 dip < nsqn rt2 dip\<close> show "False"
+        by simp
+    qed
+  qed
+
+lemma rt_strictly_fresher_eqI [intro]:
+  assumes "i\<in>kD(rt1)"
+      and "i\<in>kD(rt2)"
+      and "nsqn rt1 i = nsqn rt2 i"
+      and "\<pi>\<^sub>5(the (rt2 i)) < \<pi>\<^sub>5(the (rt1 i))"
+    shows "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2"
+  using assms unfolding rt_strictly_fresher_def' by (auto simp add: kD_nsqn)
+
+lemma invalidate_rtsf_left [simp]:
+  "\<And>dests dip rt rt'. dests dip = None \<Longrightarrow> (invalidate rt dests \<sqsubset>\<^bsub>dip\<^esub> rt') = (rt \<sqsubset>\<^bsub>dip\<^esub> rt')"
+  unfolding invalidate_def rt_strictly_fresher_def'
+  by (rule iffI) (auto split: option.split_asm)
+
+lemma vD_invalidate_rt_strictly_fresher [simp]:
+  assumes "dip \<in> vD(invalidate rt1 dests)"
+    shows "(invalidate rt1 dests \<sqsubset>\<^bsub>dip\<^esub> rt2) = (rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2)"
+  proof (cases "dip \<in> dom(dests)")
+    assume "dip \<in> dom(dests)"
+    hence "dip \<notin> vD(invalidate rt1 dests)"
+      unfolding invalidate_def vD_def
+      by clarsimp (metis assms option.simps(3) vD_invalidate_vD_not_dests)
+    with \<open>dip \<in> vD(invalidate rt1 dests)\<close> show ?thesis by simp
+  next
+    assume "dip \<notin> dom(dests)"
+    hence "dests dip = None" by auto
+    moreover with \<open>dip \<in> vD(invalidate rt1 dests)\<close> have "dip \<in> vD(rt1)"
+      unfolding invalidate_def vD_def
+      by clarsimp (metis (opaque_lifting, no_types) assms vD_Some vD_invalidate_vD_not_dests)
+    ultimately show ?thesis
+      unfolding invalidate_def rt_strictly_fresher_def' by auto
+  qed
+
+lemma rt_strictly_fresher_update_other [elim!]:
+  "\<And>dip ip rt r rt'. \<lbrakk> dip \<noteq> ip; rt \<sqsubset>\<^bsub>dip\<^esub> rt' \<rbrakk> \<Longrightarrow> update rt ip r \<sqsubset>\<^bsub>dip\<^esub> rt'"
+  unfolding rt_strictly_fresher_def' by clarsimp
+
+lemma addpreRT_strictly_fresher [simp]:
+  assumes "dip \<in> kD(rt)"
+    shows "(the (addpreRT rt dip npre) \<sqsubset>\<^bsub>ip\<^esub> rt2) = (rt \<sqsubset>\<^bsub>ip\<^esub> rt2)"
+  using assms unfolding rt_strictly_fresher_def' by clarsimp
+
+lemma lt_sqn_imp_update_strictly_fresher:
+  assumes "dip \<in> vD (rt2 nhip)"
+      and  *: "osn < sqn (rt2 nhip) dip"
+      and **: "rt \<noteq> update rt dip (osn, kno, val, hops, nhip, {})"
+    shows "update rt dip (osn, kno, val, hops, nhip, {}) \<sqsubset>\<^bsub>dip\<^esub> rt2 nhip"
+  unfolding rt_strictly_fresher_def'
+  proof (rule disjI1)
+    from ** have "nsqn (update rt dip (osn, kno, val, hops, nhip, {})) dip = osn"
+      by (rule nsqn_update_changed_kno_val)
+    with \<open>dip\<in>vD(rt2 nhip)\<close>
+      have "nsqn\<^sub>r (the (update rt dip (osn, kno, val, hops, nhip, {}) dip)) = osn"
+        by (simp add: kD_nsqn)
+    also have "osn < sqn (rt2 nhip) dip" by (rule *)
+    also have "sqn (rt2 nhip) dip = nsqn\<^sub>r (the (rt2 nhip dip))"
+      unfolding nsqn\<^sub>r_def using \<open>dip \<in> vD (rt2 nhip)\<close>
+      by - (metis vD_flag_val proj2_eq_sqn proj4_eq_flag vD_iD_gives_kD(1))
+    finally show "nsqn\<^sub>r (the (update rt dip (osn, kno, val, hops, nhip, {}) dip))
+                                                              < nsqn\<^sub>r (the (rt2 nhip dip))" .
+  qed
+
+lemma dhops_le_hops_imp_update_strictly_fresher:
+  assumes "dip \<in> vD(rt2 nhip)"
+      and sqn: "sqn (rt2 nhip) dip = osn"
+      and hop: "the (dhops (rt2 nhip) dip) \<le> hops"
+      and **: "rt \<noteq> update rt dip (osn, kno, val, Suc hops, nhip, {})"
+    shows "update rt dip (osn, kno, val, Suc hops, nhip, {}) \<sqsubset>\<^bsub>dip\<^esub> rt2 nhip"
+  unfolding rt_strictly_fresher_def'
+  proof (rule disjI2, rule conjI)
+    from ** have "nsqn (update rt dip (osn, kno, val, Suc hops, nhip, {})) dip = osn"
+      by (rule nsqn_update_changed_kno_val)
+    with \<open>dip\<in>vD(rt2 nhip)\<close>
+      have "nsqn\<^sub>r (the (update rt dip (osn, kno, val, Suc hops, nhip, {}) dip)) = osn"
+        by (simp add: kD_nsqn)
+    also have "osn = sqn (rt2 nhip) dip" by (rule sqn [symmetric])
+    also have "sqn (rt2 nhip) dip = nsqn\<^sub>r (the (rt2 nhip dip))"
+      unfolding nsqn\<^sub>r_def using \<open>dip \<in> vD(rt2 nhip)\<close>
+      by - (metis vD_flag_val proj2_eq_sqn proj4_eq_flag vD_iD_gives_kD(1))
+    finally show "nsqn\<^sub>r (the (update rt dip (osn, kno, val, Suc hops, nhip, {}) dip))
+                                                              = nsqn\<^sub>r (the (rt2 nhip dip))" .
+  next
+    have "the (dhops (rt2 nhip) dip) \<le> hops" by (rule hop)
+    also have "hops < hops + 1" by simp
+    also have "hops + 1 = the (dhops (update rt dip (osn, kno, val, Suc hops, nhip, {})) dip)"
+      using ** by simp
+    finally have "the (dhops (rt2 nhip) dip)
+                        < the (dhops (update rt dip (osn, kno, val, Suc hops, nhip, {})) dip)" .
+    thus "\<pi>\<^sub>5 (the (rt2 nhip dip)) < \<pi>\<^sub>5 (the (update rt dip (osn, kno, val, Suc hops, nhip, {}) dip))"
+      using \<open>dip \<in> vD(rt2 nhip)\<close> by (simp add: proj5_eq_dhops)
+  qed
+
+lemma nsqn_invalidate:
+  assumes "dip \<in> kD(rt)"
+      and "\<forall>ip\<in>dom(dests). ip \<in> vD(rt) \<and> the (dests ip) = inc (sqn rt ip)"
+    shows "nsqn (invalidate rt dests) dip = nsqn rt dip"
+  proof -
+    from \<open>dip \<in> kD(rt)\<close> have "dip \<in> kD(invalidate rt dests)" by simp
+
+    from assms have "rt \<approx>\<^bsub>dip\<^esub> invalidate rt dests"
+      by (rule rt_fresh_as_inc_invalidate)
+    with \<open>dip \<in> kD(rt)\<close> \<open>dip \<in> kD(invalidate rt dests)\<close> show ?thesis
+      by (simp add: kD_nsqn del: invalidate_kD_inv)
+         (erule(2) rt_fresh_as_nsqnr)
+  qed
+
+end

formal/afp/AODV/variants/a_norreqid/A_Global_Invariants.thy

@@ -0,0 +1,1159 @@
+(*  Title:       variants/a_norreqid/Global_Invariants.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "Global invariant proofs over sequential processes"
+
+theory A_Global_Invariants
+imports A_Seq_Invariants
+        A_Aodv_Predicates
+        A_Fresher
+        A_Quality_Increases
+        AWN.OAWN_Convert
+        A_OAodv
+begin
+
+lemma other_quality_increases [elim]:
+  assumes "other quality_increases I \<sigma> \<sigma>'"
+    shows "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+  using assms by (rule, clarsimp) (metis quality_increases_refl)
+
+lemma weaken_otherwith [elim]:
+    fixes m
+  assumes *: "otherwith P I (orecvmsg Q) \<sigma> \<sigma>' a"
+      and weakenP: "\<And>\<sigma> m. P \<sigma> m \<Longrightarrow> P' \<sigma> m"
+      and weakenQ: "\<And>\<sigma> m. Q \<sigma> m \<Longrightarrow> Q' \<sigma> m"
+    shows "otherwith P' I (orecvmsg Q') \<sigma> \<sigma>' a"
+  proof
+    fix j
+    assume "j\<notin>I"
+    with * have "P (\<sigma> j) (\<sigma>' j)" by auto
+    thus "P' (\<sigma> j) (\<sigma>' j)" by (rule weakenP)
+  next
+    from * have "orecvmsg Q \<sigma> a" by auto
+    thus "orecvmsg Q' \<sigma> a"
+      by rule (erule weakenQ)
+  qed
+
+lemma oreceived_msg_inv:
+  assumes other: "\<And>\<sigma> \<sigma>' m. \<lbrakk> P \<sigma> m; other Q {i} \<sigma> \<sigma>' \<rbrakk> \<Longrightarrow> P \<sigma>' m"
+      and local: "\<And>\<sigma> m. P \<sigma> m \<Longrightarrow> P (\<sigma>(i := \<sigma> i\<lparr>msg := m\<rparr>)) m"
+    shows "opaodv i \<Turnstile> (otherwith Q {i} (orecvmsg P), other Q {i} \<rightarrow>)
+                       onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<sigma>, l). l \<in> {PAodv-:1} \<longrightarrow> P \<sigma> (msg (\<sigma> i)))"
+  proof (inv_cterms, intro impI)
+    fix \<sigma> \<sigma>' l
+    assume "l = PAodv-:1 \<longrightarrow> P \<sigma> (msg (\<sigma> i))"
+       and "l = PAodv-:1"
+       and "other Q {i} \<sigma> \<sigma>'"
+    from this(1-2) have "P \<sigma> (msg (\<sigma> i))" ..
+    hence "P \<sigma>' (msg (\<sigma> i))" using \<open>other Q {i} \<sigma> \<sigma>'\<close>
+      by (rule other)
+    moreover from \<open>other Q {i} \<sigma> \<sigma>'\<close> have "\<sigma>' i = \<sigma> i" ..
+    ultimately show "P \<sigma>' (msg (\<sigma>' i))" by simp
+  next
+    fix \<sigma> \<sigma>' msg
+    assume "otherwith Q {i} (orecvmsg P) \<sigma> \<sigma>' (receive msg)"
+       and "\<sigma>' i = \<sigma> i\<lparr>msg := msg\<rparr>"
+    from this(1) have "P \<sigma> msg"
+                 and "\<forall>j. j\<noteq>i \<longrightarrow> Q (\<sigma> j) (\<sigma>' j)" by auto
+    from this(1) have "P (\<sigma>(i := \<sigma> i\<lparr>msg := msg\<rparr>)) msg" by (rule local)
+    thus "P \<sigma>' msg"
+    proof (rule other)
+      from \<open>\<sigma>' i = \<sigma> i\<lparr>msg := msg\<rparr>\<close> and \<open>\<forall>j. j\<noteq>i \<longrightarrow> Q (\<sigma> j) (\<sigma>' j)\<close>
+        show "other Q {i} (\<sigma>(i := \<sigma> i\<lparr>msg := msg\<rparr>)) \<sigma>'"
+          by - (rule otherI, auto)
+    qed
+  qed
+
+text \<open>(Equivalent to) Proposition 7.27\<close>
+
+lemma local_quality_increases:
+  "paodv i \<TTurnstile>\<^sub>A (recvmsg rreq_rrep_sn \<rightarrow>) onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). quality_increases \<xi> \<xi>')"
+  proof (rule step_invariantI)
+    fix s a s'
+    assume sr: "s \<in> reachable (paodv i) (recvmsg rreq_rrep_sn)"
+       and tr: "(s, a, s') \<in> trans (paodv i)"
+       and rm: "recvmsg rreq_rrep_sn a"
+    from sr have srTT: "s \<in> reachable (paodv i) TT" ..
+
+    from route_tables_fresher sr tr rm
+      have "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). \<forall>dip\<in>kD (rt \<xi>). rt \<xi> \<sqsubseteq>\<^bsub>dip\<^esub> rt \<xi>') (s, a, s')"
+        by (rule step_invariantD)
+
+    moreover from known_destinations_increase srTT tr TT_True
+      have "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). kD (rt \<xi>) \<subseteq> kD (rt \<xi>')) (s, a, s')"
+        by (rule step_invariantD)
+
+    moreover from sqns_increase srTT tr TT_True
+      have "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). \<forall>ip. sqn (rt \<xi>) ip \<le> sqn (rt \<xi>') ip) (s, a, s')"
+        by (rule step_invariantD)
+
+     ultimately show "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). quality_increases \<xi> \<xi>') (s, a, s')"
+       unfolding onll_def by auto
+  qed
+
+lemmas olocal_quality_increases =
+   open_seq_step_invariant [OF local_quality_increases initiali_aodv oaodv_trans aodv_trans,
+                            simplified seqll_onll_swap]
+
+lemma oquality_increases:
+  "opaodv i \<Turnstile>\<^sub>A (otherwith quality_increases {i} (orecvmsg (\<lambda>_. rreq_rrep_sn)),
+                other quality_increases {i} \<rightarrow>)
+                onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<sigma>, _), _, (\<sigma>', _)). \<forall>j. quality_increases (\<sigma> j) (\<sigma>' j))"
+  (is "_ \<Turnstile>\<^sub>A (?S, _ \<rightarrow>) _")
+  proof (rule onll_ostep_invariantI, simp)
+    fix \<sigma> p l a \<sigma>' p' l'
+    assume or: "(\<sigma>, p) \<in> oreachable (opaodv i) ?S (other quality_increases {i})"
+       and ll: "l \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p"
+       and "?S \<sigma> \<sigma>' a"
+       and tr: "((\<sigma>, p), a, (\<sigma>', p')) \<in> oseqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i"
+       and ll': "l' \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p'"
+    from this(1-3) have "orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma> a"
+      by (auto dest!: oreachable_weakenE [where QS="act (recvmsg rreq_rrep_sn)"
+                                            and QU="other quality_increases {i}"]
+                      otherwith_actionD)
+    with or have orw: "(\<sigma>, p) \<in> oreachable (opaodv i) (act (recvmsg rreq_rrep_sn))
+                                                      (other quality_increases {i})"
+      by - (erule oreachable_weakenE, auto)
+    with tr ll ll' and \<open>orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma> a\<close> have "quality_increases (\<sigma> i) (\<sigma>' i)"
+      by - (drule onll_ostep_invariantD [OF olocal_quality_increases], auto simp: seqll_def)
+    with \<open>?S \<sigma> \<sigma>' a\<close> show "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+      by (auto dest!: otherwith_syncD)
+  qed
+
+lemma rreq_rrep_nsqn_fresh_any_step_invariant:
+  "opaodv i \<Turnstile>\<^sub>A (act (recvmsg rreq_rrep_sn), other A {i} \<rightarrow>)
+                onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<sigma>, _), a, _). anycast (msg_fresh \<sigma>) a)"
+  proof (rule ostep_invariantI, simp del: act_simp)
+    fix \<sigma> p a \<sigma>' p'
+    assume or: "(\<sigma>, p) \<in> oreachable (opaodv i) (act (recvmsg rreq_rrep_sn)) (other A {i})"
+       and "((\<sigma>, p), a, (\<sigma>', p')) \<in> oseqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i"
+       and recv: "act (recvmsg rreq_rrep_sn) \<sigma> \<sigma>' a"
+    obtain l l' where "l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p" and "l'\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p'"
+      by (metis aodv_ex_label)
+    from \<open>((\<sigma>, p), a, (\<sigma>', p')) \<in> oseqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i\<close>
+      have tr: "((\<sigma>, p), a, (\<sigma>', p')) \<in> trans (opaodv i)" by simp
+
+    have "anycast (rreq_rrep_fresh (rt (\<sigma> i))) a"
+    proof -
+      have "opaodv i \<Turnstile>\<^sub>A (act (recvmsg rreq_rrep_sn), other A {i} \<rightarrow>)
+                           onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (seqll i (\<lambda>((\<xi>, _), a, _). anycast (rreq_rrep_fresh (rt \<xi>)) a))"
+        by (rule ostep_invariant_weakenE [OF
+              open_seq_step_invariant [OF rreq_rrep_fresh_any_step_invariant initiali_aodv,
+                                       simplified seqll_onll_swap]]) auto
+      hence "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (seqll i (\<lambda>((\<xi>, _), a, _). anycast (rreq_rrep_fresh (rt \<xi>)) a))
+                                                                     ((\<sigma>, p), a, (\<sigma>', p'))"
+        using or tr recv by - (erule(4) ostep_invariantE)
+      thus ?thesis
+        using \<open>l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p\<close> and \<open>l'\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p'\<close> by auto
+    qed
+
+    moreover have "anycast (rerr_invalid (rt (\<sigma> i))) a"
+    proof -
+      have "opaodv i \<Turnstile>\<^sub>A (act (recvmsg rreq_rrep_sn), other A {i} \<rightarrow>)
+                           onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (seqll i (\<lambda>((\<xi>, _), a, _). anycast (rerr_invalid (rt \<xi>)) a))"
+        by (rule ostep_invariant_weakenE [OF
+              open_seq_step_invariant [OF rerr_invalid_any_step_invariant initiali_aodv,
+                                       simplified seqll_onll_swap]]) auto
+      hence "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (seqll i (\<lambda>((\<xi>, _), a, _). anycast (rerr_invalid (rt \<xi>)) a))
+                                                                     ((\<sigma>, p), a, (\<sigma>', p'))"
+        using or tr recv by - (erule(4) ostep_invariantE)
+      thus ?thesis
+        using \<open>l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p\<close> and \<open>l'\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p'\<close> by auto
+    qed
+
+    moreover have "anycast rreq_rrep_sn a"
+    proof -
+      from or tr recv
+        have "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (seqll i (\<lambda>(_, a, _). anycast rreq_rrep_sn a)) ((\<sigma>, p), a, (\<sigma>', p'))"
+          by (rule ostep_invariantE [OF
+                open_seq_step_invariant [OF rreq_rrep_sn_any_step_invariant initiali_aodv
+                                            oaodv_trans aodv_trans,
+                                         simplified seqll_onll_swap]])
+      thus ?thesis
+        using \<open>l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p\<close> and \<open>l'\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p'\<close> by auto
+    qed
+
+    moreover have "anycast (\<lambda>m. not_Pkt m \<longrightarrow> msg_sender m = i) a"
+      proof -
+        have "opaodv i \<Turnstile>\<^sub>A (act (recvmsg rreq_rrep_sn), other A {i} \<rightarrow>)
+               onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (seqll i (\<lambda>((\<xi>, _), a, _). anycast (\<lambda>m. not_Pkt m \<longrightarrow> msg_sender m = i) a))"
+          by (rule ostep_invariant_weakenE [OF
+                open_seq_step_invariant [OF sender_ip_valid initiali_aodv,
+                                         simplified seqll_onll_swap]]) auto
+        thus ?thesis using or tr recv \<open>l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p\<close> and \<open>l'\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p'\<close>
+          by - (drule(3) onll_ostep_invariantD, auto)
+      qed
+
+    ultimately have "anycast (msg_fresh \<sigma>) a"
+      by (simp_all add: anycast_def
+                   del: msg_fresh
+                   split: seq_action.split_asm msg.split_asm) simp_all
+    thus "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<sigma>, _), a, _). anycast (msg_fresh \<sigma>) a) ((\<sigma>, p), a, (\<sigma>', p'))"
+      by auto
+  qed
+
+lemma oreceived_rreq_rrep_nsqn_fresh_inv:
+  "opaodv i \<Turnstile> (otherwith quality_increases {i} (orecvmsg msg_fresh),
+                other quality_increases {i} \<rightarrow>)
+               onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<sigma>, l). l \<in> {PAodv-:1} \<longrightarrow> msg_fresh \<sigma> (msg (\<sigma> i)))"
+  proof (rule oreceived_msg_inv)
+    fix \<sigma> \<sigma>' m
+    assume *: "msg_fresh \<sigma> m"
+       and "other quality_increases {i} \<sigma> \<sigma>'"
+    from this(2) have "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)" ..
+    thus "msg_fresh \<sigma>' m" using * ..
+  next
+    fix \<sigma> m
+    assume "msg_fresh \<sigma> m"
+    thus "msg_fresh (\<sigma>(i := \<sigma> i\<lparr>msg := m\<rparr>)) m"
+    proof (cases m)
+      fix dests sip
+      assume "m = Rerr dests sip"
+      with \<open>msg_fresh \<sigma> m\<close> show ?thesis by auto
+    qed auto
+  qed
+
+lemma oquality_increases_nsqn_fresh:
+  "opaodv i \<Turnstile>\<^sub>A (otherwith quality_increases {i} (orecvmsg msg_fresh),
+                other quality_increases {i} \<rightarrow>)
+                onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<sigma>, _), _, (\<sigma>', _)). \<forall>j. quality_increases (\<sigma> j) (\<sigma>' j))"
+  by (rule ostep_invariant_weakenE [OF oquality_increases]) auto
+
+lemma oosn_rreq:
+  "opaodv i \<Turnstile> (otherwith quality_increases {i} (orecvmsg msg_fresh),
+                other quality_increases {i} \<rightarrow>)
+       onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (seql i (\<lambda>(\<xi>, l). l \<in> {PAodv-:4, PAodv-:5} \<union> {PRreq-:n |n. True} \<longrightarrow> 1 \<le> osn \<xi>))"
+  by (rule oinvariant_weakenE [OF open_seq_invariant [OF osn_rreq initiali_aodv]])
+     (auto simp: seql_onl_swap)
+
+lemma rreq_sip:
+  "opaodv i \<Turnstile> (otherwith quality_increases {i} (orecvmsg msg_fresh),
+                other quality_increases {i} \<rightarrow>)
+               onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<sigma>, l).
+                   (l \<in> {PAodv-:4, PAodv-:5, PRreq-:0, PRreq-:2} \<and> sip (\<sigma> i) \<noteq> oip (\<sigma> i))
+                    \<longrightarrow> oip (\<sigma> i) \<in> kD(rt (\<sigma> (sip (\<sigma> i))))
+                        \<and> nsqn (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma> i)) \<ge> osn (\<sigma> i)
+                        \<and> (nsqn (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma> i)) = osn (\<sigma> i)
+                            \<longrightarrow> (hops (\<sigma> i) \<ge> the (dhops (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma> i)))
+                                  \<or> the (flag (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma> i))) = inv)))"
+    (is "_ \<Turnstile> (?S, ?U \<rightarrow>) _")
+    proof (inv_cterms inv add: oseq_step_invariant_sterms [OF oquality_increases_nsqn_fresh
+                                                              aodv_wf oaodv_trans]
+                             onl_oinvariant_sterms [OF aodv_wf oreceived_rreq_rrep_nsqn_fresh_inv]
+                             onl_oinvariant_sterms [OF aodv_wf oosn_rreq]
+                   simp add: seqlsimp
+                   simp del: One_nat_def, rule impI)
+    fix \<sigma> \<sigma>' p l
+    assume "(\<sigma>, p) \<in> oreachable (opaodv i) ?S ?U"
+       and "l \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p"
+       and pre:
+           "(l = PAodv-:4 \<or> l = PAodv-:5 \<or> l = PRreq-:0 \<or> l = PRreq-:2) \<and> sip (\<sigma> i) \<noteq> oip (\<sigma> i)
+             \<longrightarrow> oip (\<sigma> i) \<in> kD (rt (\<sigma> (sip (\<sigma> i))))
+                 \<and> osn (\<sigma> i) \<le> nsqn (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma> i))
+                 \<and> (nsqn (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma> i)) = osn (\<sigma> i)
+                    \<longrightarrow> the (dhops (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma> i))) \<le> hops (\<sigma> i)
+                        \<or> the (flag (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma> i))) = inv)"
+       and "other quality_increases {i} \<sigma> \<sigma>'"
+       and hyp: "(l=PAodv-:4 \<or> l=PAodv-:5 \<or> l=PRreq-:0 \<or> l=PRreq-:2) \<and> sip (\<sigma>' i) \<noteq> oip (\<sigma>' i)"
+           (is "?labels \<and> sip (\<sigma>' i) \<noteq> oip (\<sigma>' i)")
+    from this(4) have "\<sigma>' i = \<sigma> i" ..
+    with hyp have hyp': "?labels \<and> sip (\<sigma> i) \<noteq> oip (\<sigma> i)" by simp
+    show "oip (\<sigma>' i) \<in> kD (rt (\<sigma>' (sip (\<sigma>' i))))
+          \<and> osn (\<sigma>' i) \<le> nsqn (rt (\<sigma>' (sip (\<sigma>' i)))) (oip (\<sigma>' i))
+          \<and> (nsqn (rt (\<sigma>' (sip (\<sigma>' i)))) (oip (\<sigma>' i)) = osn (\<sigma>' i)
+             \<longrightarrow> the (dhops (rt (\<sigma>' (sip (\<sigma>' i)))) (oip (\<sigma>' i))) \<le> hops (\<sigma>' i)
+                  \<or> the (flag (rt (\<sigma>' (sip (\<sigma>' i)))) (oip (\<sigma>' i))) = inv)"
+    proof (cases "sip (\<sigma> i) = i")
+      assume "sip (\<sigma> i) \<noteq> i"
+      from \<open>other quality_increases {i} \<sigma> \<sigma>'\<close>
+        have "quality_increases (\<sigma> (sip (\<sigma> i))) (\<sigma>' (sip (\<sigma>' i)))"
+          by (rule otherE)  (clarsimp simp: \<open>sip (\<sigma> i) \<noteq> i\<close>)
+      moreover from \<open>(\<sigma>, p) \<in> oreachable (opaodv i) ?S ?U\<close> \<open>l \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p\<close> and hyp
+        have "1 \<le> osn (\<sigma>' i)"
+          by (auto dest!: onl_oinvariant_weakenD [OF oosn_rreq]
+                   simp add: seqlsimp \<open>\<sigma>' i = \<sigma> i\<close>)
+      moreover from \<open>sip (\<sigma> i) \<noteq> i\<close> hyp' and pre
+        have "oip (\<sigma>' i) \<in> kD (rt (\<sigma> (sip (\<sigma> i))))
+              \<and> osn (\<sigma>' i) \<le> nsqn (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma>' i))
+              \<and> (nsqn (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma>' i)) = osn (\<sigma>' i)
+                  \<longrightarrow> the (dhops (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma>' i))) \<le> hops (\<sigma>' i)
+                       \<or> the (flag (rt (\<sigma> (sip (\<sigma> i)))) (oip (\<sigma>' i))) = inv)"
+        by (auto simp: \<open>\<sigma>' i = \<sigma> i\<close>)
+      ultimately show ?thesis
+        by (rule quality_increases_rreq_rrep_props)
+    next
+      assume "sip (\<sigma> i) = i" thus ?thesis
+        using \<open>\<sigma>' i = \<sigma> i\<close> hyp and pre by auto
+    qed
+  qed (auto elim!: quality_increases_rreq_rrep_props')
+
+lemma odsn_rrep:
+  "opaodv i \<Turnstile> (otherwith quality_increases {i} (orecvmsg msg_fresh),
+                other quality_increases {i} \<rightarrow>)
+      onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (seql i (\<lambda>(\<xi>, l). l \<in> {PAodv-:6, PAodv-:7} \<union> {PRrep-:n|n. True} \<longrightarrow> 1 \<le> dsn \<xi>))"
+  by (rule oinvariant_weakenE [OF open_seq_invariant [OF dsn_rrep initiali_aodv]])
+     (auto simp: seql_onl_swap)
+
+lemma rrep_sip:
+  "opaodv i \<Turnstile> (otherwith quality_increases {i} (orecvmsg msg_fresh),
+                other quality_increases {i} \<rightarrow>)
+               onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<sigma>, l).
+                   (l \<in> {PAodv-:6, PAodv-:7, PRrep-:0, PRrep-:1} \<and> sip (\<sigma> i) \<noteq> dip (\<sigma> i))
+                    \<longrightarrow> dip (\<sigma> i) \<in> kD(rt (\<sigma> (sip (\<sigma> i))))
+                        \<and> nsqn (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma> i)) \<ge> dsn (\<sigma> i)
+                        \<and> (nsqn (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma> i)) = dsn (\<sigma> i)
+                            \<longrightarrow> (hops (\<sigma> i) \<ge> the (dhops (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma> i)))
+                                  \<or> the (flag (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma> i))) = inv)))"
+    (is "_ \<Turnstile> (?S, ?U \<rightarrow>) _")
+  proof (inv_cterms inv add: oseq_step_invariant_sterms [OF oquality_increases_nsqn_fresh aodv_wf
+                                                            oaodv_trans]
+                             onl_oinvariant_sterms [OF aodv_wf oreceived_rreq_rrep_nsqn_fresh_inv]
+                             onl_oinvariant_sterms [OF aodv_wf odsn_rrep]
+                   simp del: One_nat_def, rule impI)
+    fix \<sigma> \<sigma>' p l
+    assume "(\<sigma>, p) \<in> oreachable (opaodv i) ?S ?U"
+       and "l \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p"
+       and pre:
+           "(l = PAodv-:6 \<or> l = PAodv-:7 \<or> l = PRrep-:0 \<or> l = PRrep-:1) \<and> sip (\<sigma> i) \<noteq> dip (\<sigma> i)
+           \<longrightarrow> dip (\<sigma> i) \<in> kD (rt (\<sigma> (sip (\<sigma> i))))
+               \<and> dsn (\<sigma> i) \<le> nsqn (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma> i))
+               \<and> (nsqn (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma> i)) = dsn (\<sigma> i)
+                  \<longrightarrow> the (dhops (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma> i))) \<le> hops (\<sigma> i)
+                      \<or> the (flag (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma> i))) = inv)"
+       and "other quality_increases {i} \<sigma> \<sigma>'"
+       and hyp: "(l=PAodv-:6 \<or> l=PAodv-:7 \<or> l=PRrep-:0 \<or> l=PRrep-:1) \<and> sip (\<sigma>' i) \<noteq> dip (\<sigma>' i)"
+           (is "?labels \<and> sip (\<sigma>' i) \<noteq> dip (\<sigma>' i)")
+    from this(4) have "\<sigma>' i = \<sigma> i" ..
+    with hyp have hyp': "?labels \<and> sip (\<sigma> i) \<noteq> dip (\<sigma> i)" by simp
+    show "dip (\<sigma>' i) \<in> kD (rt (\<sigma>' (sip (\<sigma>' i))))
+          \<and> dsn (\<sigma>' i) \<le> nsqn (rt (\<sigma>' (sip (\<sigma>' i)))) (dip (\<sigma>' i))
+          \<and> (nsqn (rt (\<sigma>' (sip (\<sigma>' i)))) (dip (\<sigma>' i)) = dsn (\<sigma>' i)
+             \<longrightarrow> the (dhops (rt (\<sigma>' (sip (\<sigma>' i)))) (dip (\<sigma>' i))) \<le> hops (\<sigma>' i)
+                 \<or> the (flag (rt (\<sigma>' (sip (\<sigma>' i)))) (dip (\<sigma>' i))) = inv)"
+    proof (cases "sip (\<sigma> i) = i")
+      assume "sip (\<sigma> i) \<noteq> i"
+      from \<open>other quality_increases {i} \<sigma> \<sigma>'\<close>
+        have "quality_increases (\<sigma> (sip (\<sigma> i))) (\<sigma>' (sip (\<sigma>' i)))"
+          by (rule otherE)  (clarsimp simp: \<open>sip (\<sigma> i) \<noteq> i\<close>)
+      moreover from \<open>(\<sigma>, p) \<in> oreachable (opaodv i) ?S ?U\<close> \<open>l \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p\<close> and hyp
+        have "1 \<le> dsn (\<sigma>' i)"
+          by (auto dest!: onl_oinvariant_weakenD [OF odsn_rrep]
+                   simp add: seqlsimp \<open>\<sigma>' i = \<sigma> i\<close>)
+      moreover from \<open>sip (\<sigma> i) \<noteq> i\<close> hyp' and pre
+        have "dip (\<sigma>' i) \<in> kD (rt (\<sigma> (sip (\<sigma> i))))
+              \<and> dsn (\<sigma>' i) \<le> nsqn (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma>' i))
+              \<and> (nsqn (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma>' i)) = dsn (\<sigma>' i)
+                 \<longrightarrow> the (dhops (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma>' i))) \<le> hops (\<sigma>' i)
+                     \<or> the (flag (rt (\<sigma> (sip (\<sigma> i)))) (dip (\<sigma>' i))) = inv)"
+        by (auto simp: \<open>\<sigma>' i = \<sigma> i\<close>)
+      ultimately show ?thesis
+        by (rule quality_increases_rreq_rrep_props)
+    next
+      assume "sip (\<sigma> i) = i" thus ?thesis
+        using \<open>\<sigma>' i = \<sigma> i\<close> hyp and pre by auto
+    qed
+  qed (auto simp add: seqlsimp elim!: quality_increases_rreq_rrep_props')
+
+lemma rerr_sip:
+  "opaodv i \<Turnstile> (otherwith quality_increases {i} (orecvmsg msg_fresh),
+                other quality_increases {i} \<rightarrow>)
+               onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<sigma>, l).
+                 l \<in> {PAodv-:8, PAodv-:9, PRerr-:0, PRerr-:1}
+                 \<longrightarrow> (\<forall>ripc\<in>dom(dests (\<sigma> i)). ripc\<in>kD(rt (\<sigma> (sip (\<sigma> i)))) \<and>
+                        the (dests (\<sigma> i) ripc) - 1 \<le> nsqn (rt (\<sigma> (sip (\<sigma> i)))) ripc))"
+  (is "_ \<Turnstile> (?S, ?U \<rightarrow>) _")
+  proof -
+    { fix dests rip sip rsn and \<sigma> \<sigma>' :: "ip \<Rightarrow> state"
+      assume qinc: "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+         and *: "\<forall>rip\<in>dom dests. rip \<in> kD (rt (\<sigma> sip))
+                                  \<and> the (dests rip) - 1 \<le> nsqn (rt (\<sigma> sip)) rip"
+         and "dests rip = Some rsn"
+      from this(3) have "rip\<in>dom dests" by auto
+      with * and \<open>dests rip = Some rsn\<close> have "rip\<in>kD(rt (\<sigma> sip))"
+                                         and "rsn - 1 \<le> nsqn (rt (\<sigma> sip)) rip"
+        by (auto dest!: bspec)
+      from qinc have "quality_increases (\<sigma> sip) (\<sigma>' sip)" ..
+      have "rip \<in> kD(rt (\<sigma>' sip)) \<and> rsn - 1 \<le> nsqn (rt (\<sigma>' sip)) rip"
+      proof
+        from \<open>rip\<in>kD(rt (\<sigma> sip))\<close> and \<open>quality_increases (\<sigma> sip) (\<sigma>' sip)\<close>
+          show "rip \<in> kD(rt (\<sigma>' sip))" ..
+      next
+        from \<open>rip\<in>kD(rt (\<sigma> sip))\<close> and \<open>quality_increases (\<sigma> sip) (\<sigma>' sip)\<close>
+          have "nsqn (rt (\<sigma> sip)) rip \<le> nsqn (rt (\<sigma>' sip)) rip" ..
+        with \<open>rsn - 1 \<le> nsqn (rt (\<sigma> sip)) rip\<close> show "rsn - 1 \<le> nsqn (rt (\<sigma>' sip)) rip"
+          by (rule le_trans)
+      qed
+    } note partial = this
+
+    show ?thesis
+      by (inv_cterms inv add: oseq_step_invariant_sterms [OF oquality_increases_nsqn_fresh aodv_wf
+                                                             oaodv_trans]
+                              onl_oinvariant_sterms [OF aodv_wf oreceived_rreq_rrep_nsqn_fresh_inv]
+                              other_quality_increases other_localD
+                    simp del: One_nat_def, intro conjI)
+         (clarsimp simp del: One_nat_def split: if_split_asm option.split_asm, erule(2) partial)+
+  qed
+
+lemma prerr_guard: "paodv i \<TTurnstile>
+                  onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). (l = PRerr-:1
+                      \<longrightarrow> (\<forall>ip\<in>dom(dests \<xi>). ip\<in>vD(rt \<xi>)
+                                             \<and> the (nhop (rt \<xi>) ip) = sip \<xi>
+                                             \<and> sqn (rt \<xi>) ip < the (dests \<xi> ip))))"
+  by (inv_cterms) (clarsimp split: option.split_asm if_split_asm)
+
+lemmas oaddpreRT_welldefined =
+         open_seq_invariant [OF addpreRT_welldefined initiali_aodv oaodv_trans aodv_trans,
+                             simplified seql_onl_swap,
+                             THEN oinvariant_anyact]
+
+lemmas odests_vD_inc_sqn =
+         open_seq_invariant [OF dests_vD_inc_sqn initiali_aodv oaodv_trans aodv_trans,
+                             simplified seql_onl_swap,
+                             THEN oinvariant_anyact]
+
+lemmas oprerr_guard =
+         open_seq_invariant [OF prerr_guard initiali_aodv oaodv_trans aodv_trans,
+                             simplified seql_onl_swap,
+                             THEN oinvariant_anyact]
+
+text \<open>Proposition 7.28\<close>
+
+lemma seq_compare_next_hop':
+  "opaodv i \<Turnstile> (otherwith quality_increases {i} (orecvmsg msg_fresh),
+                other quality_increases {i} \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<sigma>, _).
+                   \<forall>dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                         in dip \<in> kD(rt (\<sigma> i)) \<and> nhip \<noteq> dip \<longrightarrow>
+                            dip \<in> kD(rt (\<sigma> nhip)) \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> nhip)) dip)"
+  (is "_ \<Turnstile> (?S, ?U \<rightarrow>) _")
+  proof -
+
+  { fix nhop and \<sigma> \<sigma>' :: "ip \<Rightarrow> state"
+    assume  pre: "\<forall>dip\<in>kD(rt (\<sigma> i)). nhop dip \<noteq> dip \<longrightarrow>
+                    dip\<in>kD(rt (\<sigma> (nhop dip))) \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> (nhop dip))) dip"
+       and qinc: "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+    have "\<forall>dip\<in>kD(rt (\<sigma> i)). nhop dip \<noteq> dip \<longrightarrow>
+                  dip\<in>kD(rt (\<sigma>' (nhop dip))) \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip"
+    proof (intro ballI impI)
+      fix dip
+      assume "dip\<in>kD(rt (\<sigma> i))"
+         and "nhop dip \<noteq> dip"
+      with pre have "dip\<in>kD(rt (\<sigma> (nhop dip)))"
+                and "nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> (nhop dip))) dip"
+        by auto
+      from qinc have qinc_nhop: "quality_increases (\<sigma> (nhop dip)) (\<sigma>' (nhop dip))" ..
+      with \<open>dip\<in>kD(rt (\<sigma> (nhop dip)))\<close> have "dip\<in>kD (rt (\<sigma>' (nhop dip)))" ..
+
+      moreover have "nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip"
+      proof -
+        from \<open>dip\<in>kD(rt (\<sigma> (nhop dip)))\<close> qinc_nhop
+          have "nsqn (rt (\<sigma> (nhop dip))) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip" ..
+        with \<open>nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> (nhop dip))) dip\<close> show ?thesis
+          by simp
+      qed
+
+      ultimately show "dip\<in>kD(rt (\<sigma>' (nhop dip)))
+                       \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip" ..
+    qed
+  } note basic = this
+
+  { fix nhop and \<sigma> \<sigma>' :: "ip \<Rightarrow> state"
+    assume pre: "\<forall>dip\<in>kD(rt (\<sigma> i)). nhop dip \<noteq> dip \<longrightarrow> dip\<in>kD(rt (\<sigma> (nhop dip)))
+                                             \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> (nhop dip))) dip"
+       and ndest: "\<forall>ripc\<in>dom (dests (\<sigma> i)). ripc \<in> kD (rt (\<sigma> (sip (\<sigma> i))))
+                                   \<and> the (dests (\<sigma> i) ripc) - 1 \<le> nsqn (rt (\<sigma> (sip (\<sigma> i)))) ripc"
+       and issip: "\<forall>ip\<in>dom (dests (\<sigma> i)). nhop ip = sip (\<sigma> i)"
+       and qinc: "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+    have "\<forall>dip\<in>kD(rt (\<sigma> i)). nhop dip \<noteq> dip \<longrightarrow> dip \<in> kD (rt (\<sigma>' (nhop dip)))
+                 \<and> nsqn (invalidate (rt (\<sigma> i)) (dests (\<sigma> i))) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip"
+    proof (intro ballI impI)
+      fix dip
+      assume "dip\<in>kD(rt (\<sigma> i))"
+         and "nhop dip \<noteq> dip"
+      with pre and qinc have "dip\<in>kD(rt (\<sigma>' (nhop dip)))"
+                         and "nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip"
+        by (auto dest!: basic)
+
+      have "nsqn (invalidate (rt (\<sigma> i)) (dests (\<sigma> i))) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip"
+      proof (cases "dip\<in>dom (dests (\<sigma> i))")
+        assume "dip\<in>dom (dests (\<sigma> i))"
+        with \<open>dip\<in>kD(rt (\<sigma> i))\<close> obtain dsn where "dests (\<sigma> i) dip = Some dsn"
+          by auto
+        with \<open>dip\<in>kD(rt (\<sigma> i))\<close> have "nsqn (invalidate (rt (\<sigma> i)) (dests (\<sigma> i))) dip = dsn - 1"
+           by (rule nsqn_invalidate_eq)
+        moreover have "dsn - 1 \<le> nsqn (rt (\<sigma>' (nhop dip))) dip"
+        proof -
+          from \<open>dests (\<sigma> i) dip = Some dsn\<close> have "the (dests (\<sigma> i) dip) = dsn" by simp
+          with ndest and \<open>dip\<in>dom (dests (\<sigma> i))\<close> have "dip \<in> kD (rt (\<sigma> (sip (\<sigma> i))))"
+                                                      "dsn - 1 \<le> nsqn (rt (\<sigma> (sip (\<sigma> i)))) dip"
+            by auto
+          moreover from issip and \<open>dip\<in>dom (dests (\<sigma> i))\<close> have "nhop dip = sip (\<sigma> i)" ..
+          ultimately have "dip \<in> kD (rt (\<sigma> (nhop dip)))"
+                      and "dsn - 1 \<le> nsqn (rt (\<sigma> (nhop dip))) dip" by auto
+          with qinc show "dsn - 1 \<le> nsqn (rt (\<sigma>' (nhop dip))) dip"
+            by simp (metis kD_nsqn_quality_increases_trans)
+        qed
+        ultimately show ?thesis by simp
+      next
+        assume "dip \<notin> dom (dests (\<sigma> i))"
+        with \<open>dip\<in>kD(rt (\<sigma> i))\<close>
+          have "nsqn (invalidate (rt (\<sigma> i)) (dests (\<sigma> i))) dip = nsqn (rt (\<sigma> i)) dip"
+            by (rule nsqn_invalidate_other)
+        with \<open>nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip\<close> show ?thesis by simp
+      qed
+      with \<open>dip\<in>kD(rt (\<sigma>' (nhop dip)))\<close>
+        show "dip \<in> kD (rt (\<sigma>' (nhop dip)))
+              \<and> nsqn (invalidate (rt (\<sigma> i)) (dests (\<sigma> i))) dip \<le> nsqn (rt (\<sigma>' (nhop dip))) dip" ..
+    qed
+  } note basic_prerr = this
+
+  { fix \<sigma> \<sigma>' :: "ip \<Rightarrow> state"
+    assume a1: "\<forall>dip\<in>kD(rt (\<sigma> i)). the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                \<longrightarrow> dip\<in>kD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                    \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))) dip"
+       and a2: "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+    have "\<forall>dip\<in>kD(rt (\<sigma> i)).
+          the (nhop (update (rt (\<sigma> i)) (sip (\<sigma> i)) (0, unk, val, Suc 0, sip (\<sigma> i), {})) dip) \<noteq> dip \<longrightarrow>
+          dip\<in>kD(rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) (sip (\<sigma> i))
+                                        (0, unk, val, Suc 0, sip (\<sigma> i), {}))
+                                  dip)))) \<and>
+          nsqn (update (rt (\<sigma> i)) (sip (\<sigma> i)) (0, unk, val, Suc 0, sip (\<sigma> i), {})) dip
+          \<le> nsqn (rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) (sip (\<sigma> i))
+                                      (0, unk, val, Suc 0, sip (\<sigma> i), {}))
+                                dip))))
+             dip" (is "\<forall>dip\<in>kD(rt (\<sigma> i)). ?P dip")
+    proof
+      fix dip
+      assume "dip\<in>kD(rt (\<sigma> i))"
+      with a1 and a2  
+        have "the (nhop (rt (\<sigma> i)) dip) \<noteq> dip \<longrightarrow> dip\<in>kD(rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip))))
+                        \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip)))) dip"
+           by - (drule(1) basic, auto)
+      thus "?P dip" by (cases "dip = sip (\<sigma> i)") auto
+    qed
+  } note nhop_update_sip = this
+
+  { fix \<sigma> \<sigma>' oip sip osn hops
+    assume pre: "\<forall>dip\<in>kD (rt (\<sigma> i)). the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                 \<longrightarrow> dip\<in>kD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                     \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))) dip"
+       and qinc: "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+       and *: "sip \<noteq> oip \<longrightarrow> oip\<in>kD(rt (\<sigma> sip))
+                             \<and> osn \<le> nsqn (rt (\<sigma> sip)) oip
+                             \<and> (nsqn (rt (\<sigma> sip)) oip = osn
+                                \<longrightarrow> the (dhops (rt (\<sigma> sip)) oip) \<le> hops
+                                    \<or> the (flag (rt (\<sigma> sip)) oip) = inv)"
+    from pre and qinc
+      have pre': "\<forall>dip\<in>kD (rt (\<sigma> i)). the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                   \<longrightarrow> dip\<in>kD(rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip))))
+                       \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip)))) dip"
+        by (rule basic)
+    have "(the (nhop (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})) oip) \<noteq> oip
+           \<longrightarrow> oip\<in>kD(rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) oip
+                                                   (osn, kno, val, Suc hops, sip, {})) oip))))
+                \<and> nsqn (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})) oip
+                   \<le> nsqn (rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) oip
+                                                   (osn, kno, val, Suc hops, sip, {})) oip)))) oip)"
+       (is "?nhop_not_oip \<longrightarrow> ?oip_in_kD \<and> ?nsqn_le_nsqn")
+     proof (rule, split update_rt_split_asm)
+       assume "rt (\<sigma> i) = update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})"
+          and "the (nhop (rt (\<sigma> i)) oip) \<noteq> oip"
+       with pre' show "?oip_in_kD \<and> ?nsqn_le_nsqn" by auto
+     next
+       assume rtnot: "rt (\<sigma> i) \<noteq> update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})"
+          and notoip: ?nhop_not_oip
+       with * qinc have ?oip_in_kD
+         by (clarsimp elim!: kD_quality_increases)
+       moreover with * pre qinc rtnot notoip have ?nsqn_le_nsqn
+        by simp (metis kD_nsqn_quality_increases_trans)
+       ultimately show "?oip_in_kD \<and> ?nsqn_le_nsqn" ..
+     qed
+  } note update1 = this
+
+  { fix \<sigma> \<sigma>' oip sip osn hops
+    assume pre: "\<forall>dip\<in>kD (rt (\<sigma> i)). the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                  \<longrightarrow> dip\<in>kD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                      \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))) dip"
+       and qinc: "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+       and *: "sip \<noteq> oip \<longrightarrow> oip\<in>kD(rt (\<sigma> sip))
+                           \<and> osn \<le> nsqn (rt (\<sigma> sip)) oip
+                           \<and> (nsqn (rt (\<sigma> sip)) oip = osn
+                              \<longrightarrow> the (dhops (rt (\<sigma> sip)) oip) \<le> hops
+                                  \<or> the (flag (rt (\<sigma> sip)) oip) = inv)"
+    from pre and qinc
+      have pre': "\<forall>dip\<in>kD (rt (\<sigma> i)). the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                   \<longrightarrow> dip\<in>kD(rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip))))
+                       \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip)))) dip"
+        by (rule basic)
+    have "\<forall>dip\<in>kD(rt (\<sigma> i)).
+           the (nhop (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})) dip) \<noteq> dip
+           \<longrightarrow> dip\<in>kD(rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) oip
+                                                  (osn, kno, val, Suc hops, sip, {})) dip))))
+               \<and> nsqn (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})) dip
+                  \<le> nsqn (rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) oip
+                                                  (osn, kno, val, Suc hops, sip, {})) dip)))) dip"
+       (is "\<forall>dip\<in>kD(rt (\<sigma> i)). _ \<longrightarrow> ?dip_in_kD dip \<and> ?nsqn_le_nsqn dip")
+     proof (intro ballI impI, split update_rt_split_asm)
+       fix dip
+       assume "dip\<in>kD(rt (\<sigma> i))"
+          and "the (nhop (rt (\<sigma> i)) dip) \<noteq> dip"
+          and "rt (\<sigma> i) = update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})"
+        with pre' show "?dip_in_kD dip \<and> ?nsqn_le_nsqn dip" by simp
+     next
+       fix dip
+       assume "dip\<in>kD(rt (\<sigma> i))"
+          and notdip: "the (nhop (update (rt (\<sigma> i)) oip
+                             (osn, kno, val, Suc hops, sip, {})) dip) \<noteq> dip"
+          and rtnot: "rt (\<sigma> i) \<noteq> update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})"
+       show "?dip_in_kD dip \<and> ?nsqn_le_nsqn dip"
+       proof (cases "dip = oip")
+         assume "dip \<noteq> oip"
+         with pre' \<open>dip\<in>kD(rt (\<sigma> i))\<close> notdip
+           show ?thesis by clarsimp
+       next
+         assume "dip = oip"
+         with rtnot qinc \<open>dip\<in>kD(rt (\<sigma> i))\<close> notdip *
+           have "?dip_in_kD dip"
+            by simp (metis kD_quality_increases)
+         moreover from \<open>dip = oip\<close> rtnot qinc \<open>dip\<in>kD(rt (\<sigma> i))\<close> notdip *
+           have "?nsqn_le_nsqn dip" by simp (metis kD_nsqn_quality_increases_trans)
+         ultimately show ?thesis ..
+       qed
+     qed
+  } note update2 = this
+
+  have "opaodv i \<Turnstile> (?S, ?U \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<sigma>, _).
+                   \<forall>dip \<in> kD(rt (\<sigma> i)). the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                      \<longrightarrow> dip \<in> kD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                          \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))) dip)"
+    by (inv_cterms inv add: oseq_step_invariant_sterms [OF oquality_increases_nsqn_fresh aodv_wf
+                                                           oaodv_trans]
+                            onl_oinvariant_sterms [OF aodv_wf oaddpreRT_welldefined]
+                            onl_oinvariant_sterms [OF aodv_wf odests_vD_inc_sqn]
+                            onl_oinvariant_sterms [OF aodv_wf oprerr_guard]
+                            onl_oinvariant_sterms [OF aodv_wf rreq_sip]
+                            onl_oinvariant_sterms [OF aodv_wf rrep_sip]
+                            onl_oinvariant_sterms [OF aodv_wf rerr_sip]
+                            other_quality_increases
+                            other_localD
+                      solve: basic basic_prerr
+                      simp add: seqlsimp nsqn_invalidate nhop_update_sip
+                      simp del: One_nat_def)
+       (rule conjI, erule(2) update1, erule(2) update2)+
+
+    thus ?thesis unfolding Let_def by auto
+  qed
+
+text \<open>Proposition 7.30\<close>
+
+lemmas okD_unk_or_atleast_one =
+         open_seq_invariant [OF kD_unk_or_atleast_one initiali_aodv,
+                             simplified seql_onl_swap]
+
+lemmas ozero_seq_unk_hops_one =
+         open_seq_invariant [OF zero_seq_unk_hops_one initiali_aodv,
+                             simplified seql_onl_swap]
+
+lemma oreachable_fresh_okD_unk_or_atleast_one:
+    fixes dip
+  assumes "(\<sigma>, p) \<in> oreachable (opaodv i)
+                       (otherwith ((=)) {i} (orecvmsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m
+                                                             \<and> msg_zhops m)))
+                       (other quality_increases {i})"
+      and "dip\<in>kD(rt (\<sigma> i))"
+    shows "\<pi>\<^sub>3(the (rt (\<sigma> i) dip)) = unk \<or> 1 \<le> \<pi>\<^sub>2(the (rt (\<sigma> i) dip))"
+    (is "?P dip")
+  proof -
+    have "\<exists>l. l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p" by (metis aodv_ex_label)
+    with assms(1) have "\<forall>dip\<in>kD (rt (\<sigma> i)). ?P dip"
+      by - (drule oinvariant_weakenD [OF okD_unk_or_atleast_one [OF oaodv_trans aodv_trans]],
+            auto dest!: otherwith_actionD onlD simp: seqlsimp)
+    with \<open>dip\<in>kD(rt (\<sigma> i))\<close> show ?thesis by simp
+  qed
+
+lemma oreachable_fresh_ozero_seq_unk_hops_one:
+    fixes dip
+  assumes "(\<sigma>, p) \<in> oreachable (opaodv i)
+                     (otherwith ((=)) {i} (orecvmsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m
+                                                         \<and> msg_zhops m)))
+                     (other quality_increases {i})"
+      and "dip\<in>kD(rt (\<sigma> i))"
+    shows "sqn (rt (\<sigma> i)) dip = 0 \<longrightarrow>
+             sqnf (rt (\<sigma> i)) dip = unk
+             \<and> the (dhops (rt (\<sigma> i)) dip) = 1
+             \<and> the (nhop (rt (\<sigma> i)) dip) = dip"
+    (is "?P dip")
+  proof -
+    have "\<exists>l. l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p" by (metis aodv_ex_label)
+    with assms(1) have "\<forall>dip\<in>kD (rt (\<sigma> i)). ?P dip"
+      by - (drule oinvariant_weakenD [OF ozero_seq_unk_hops_one [OF oaodv_trans aodv_trans]],
+            auto dest!: onlD otherwith_actionD simp: seqlsimp)
+    with \<open>dip\<in>kD(rt (\<sigma> i))\<close> show ?thesis by simp
+  qed
+
+lemma seq_nhop_quality_increases':
+  shows "opaodv i \<Turnstile> (otherwith ((=)) {i}
+                        (orecvmsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)),
+                         other quality_increases {i} \<rightarrow>)
+                        onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<sigma>, _). \<forall>dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                               in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip))
+                                                  \<and> nhip \<noteq> dip
+                                                  \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+  (is "_ \<Turnstile> (?S i, _ \<rightarrow>) _")
+  proof -
+    have weaken:
+      "\<And>p I Q R P. p \<Turnstile> (otherwith quality_increases I (orecvmsg Q), other quality_increases I \<rightarrow>) P
+       \<Longrightarrow> p \<Turnstile> (otherwith ((=)) I (orecvmsg (\<lambda>\<sigma> m. Q \<sigma> m \<and> R \<sigma> m)), other quality_increases I \<rightarrow>) P"
+        by auto
+    {
+      fix i a and \<sigma> \<sigma>' :: "ip \<Rightarrow> state"
+      assume a1: "\<forall>dip. dip\<in>vD(rt (\<sigma> i))
+                        \<and> dip\<in>vD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                        \<and> (the (nhop (rt (\<sigma> i)) dip)) \<noteq> dip
+                         \<longrightarrow> rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))"
+         and ow: "?S i \<sigma> \<sigma>' a"
+      have "\<forall>dip. dip\<in>vD(rt (\<sigma> i))
+                  \<and> dip\<in>vD (rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip))))
+                  \<and> (the (nhop (rt (\<sigma> i)) dip)) \<noteq> dip
+               \<longrightarrow> rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip)))"
+      proof clarify
+        fix dip
+        assume a2: "dip\<in>vD(rt (\<sigma> i))"
+           and a3: "dip\<in>vD (rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip))))"
+           and a4: "(the (nhop (rt (\<sigma> i)) dip)) \<noteq> dip"
+        from ow have "\<forall>j. j \<noteq> i \<longrightarrow> \<sigma> j = \<sigma>' j" by auto
+        show "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip)))"
+        proof (cases "(the (nhop (rt (\<sigma> i)) dip)) = i")
+          assume "(the (nhop (rt (\<sigma> i)) dip)) = i"
+          with \<open>dip \<in> vD(rt (\<sigma> i))\<close> have "dip \<in> vD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))" by simp
+          with a1 a2 a4 have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))" by simp
+          with \<open>(the (nhop (rt (\<sigma> i)) dip)) = i\<close> have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> i)" by simp
+          hence False by simp
+          thus ?thesis ..
+        next
+          assume "(the (nhop (rt (\<sigma> i)) dip)) \<noteq> i"
+          with \<open>\<forall>j. j \<noteq> i \<longrightarrow> \<sigma> j = \<sigma>' j\<close>
+            have *: "\<sigma> (the (nhop (rt (\<sigma> i)) dip)) = \<sigma>' (the (nhop (rt (\<sigma> i)) dip))" by simp
+          with \<open>dip\<in>vD (rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip))))\<close>
+            have "dip\<in>vD (rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))" by simp
+          with a1 a2 a4 have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))" by simp
+          with * show ?thesis by simp
+        qed
+      qed
+    } note basic = this
+
+    { fix \<sigma> \<sigma>' a dip sip i
+      assume a1: "\<forall>dip. dip\<in>vD(rt (\<sigma> i))
+                      \<and> dip\<in>vD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                      \<and> the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                      \<longrightarrow> rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))"
+         and ow: "?S i \<sigma> \<sigma>' a"
+      have "\<forall>dip. dip\<in>vD(update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {}))
+           \<and> dip\<in>vD(rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})) dip))))
+           \<and> the (nhop (update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})) dip) \<noteq> dip
+           \<longrightarrow> update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})
+               \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})) dip)))"
+      proof clarify
+        fix dip
+        assume a2: "dip\<in>vD (update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {}))"
+           and a3: "dip\<in>vD(rt (\<sigma>' (the (nhop
+                         (update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})) dip))))"
+           and a4: "the (nhop (update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})) dip) \<noteq> dip"
+        show "update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})
+              \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})) dip)))"
+        proof (cases "dip = sip")
+          assume "dip = sip"
+          with \<open>the (nhop (update (rt (\<sigma> i)) sip (0, unk, val, Suc 0, sip, {})) dip) \<noteq> dip\<close>
+          have False by simp
+          thus ?thesis ..
+        next
+          assume [simp]: "dip \<noteq> sip"
+          from a2 have "dip\<in>vD(rt (\<sigma> i)) \<or> dip = sip"
+            by (rule vD_update_val)
+          with \<open>dip \<noteq> sip\<close> have "dip\<in>vD(rt (\<sigma> i))" by simp
+          moreover from a3 have "dip\<in>vD(rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip))))" by simp
+          moreover from a4 have "the (nhop (rt (\<sigma> i)) dip) \<noteq> dip" by simp
+          ultimately have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip)))"
+            using a1 ow by - (drule(1) basic, simp)
+          with \<open>dip \<noteq> sip\<close> show ?thesis
+            by - (erule rt_strictly_fresher_update_other, simp)
+        qed
+      qed
+    } note update_0_unk = this
+
+    { fix \<sigma> a \<sigma>' nhop
+      assume pre: "\<forall>dip. dip\<in>vD(rt (\<sigma> i)) \<and> dip\<in>vD(rt (\<sigma> (nhop dip))) \<and> nhop dip \<noteq> dip
+                         \<longrightarrow> rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (nhop dip))"
+         and ow: "?S i \<sigma> \<sigma>' a"
+      have "\<forall>dip. dip \<in> vD (invalidate (rt (\<sigma> i)) (dests (\<sigma> i)))
+                  \<and> dip \<in> vD (rt (\<sigma>' (nhop dip))) \<and> nhop dip \<noteq> dip
+                  \<longrightarrow> rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (nhop dip))"
+      proof clarify
+        fix dip
+        assume "dip\<in>vD(invalidate (rt (\<sigma> i)) (dests (\<sigma> i)))"
+           and "dip\<in>vD(rt (\<sigma>' (nhop dip)))"
+           and "nhop dip \<noteq> dip"
+        from this(1) have "dip\<in>vD (rt (\<sigma> i))"
+          by (clarsimp dest!: vD_invalidate_vD_not_dests)
+        moreover from ow have "\<forall>j. j \<noteq> i \<longrightarrow> \<sigma> j = \<sigma>' j" by auto
+        ultimately have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (nhop dip))"
+          using pre \<open>dip \<in> vD (rt (\<sigma>' (nhop dip)))\<close> \<open>nhop dip \<noteq> dip\<close>
+          by metis
+        with \<open>\<forall>j. j \<noteq> i \<longrightarrow> \<sigma> j = \<sigma>' j\<close> show  "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (nhop dip))"
+          by (metis rt_strictly_fresher_irefl)
+      qed
+    } note invalidate = this
+
+    { fix \<sigma> a \<sigma>' dip oip osn sip hops i
+      assume pre: "\<forall>dip. dip \<in> vD (rt (\<sigma> i))
+                       \<and> dip \<in> vD (rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                       \<and> the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                   \<longrightarrow> rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))"
+         and ow: "?S i \<sigma> \<sigma>' a"
+         and "Suc 0 \<le> osn"
+         and a6: "sip \<noteq> oip \<longrightarrow> oip \<in> kD (rt (\<sigma> sip))
+                                 \<and> osn \<le> nsqn (rt (\<sigma> sip)) oip
+                                 \<and> (nsqn (rt (\<sigma> sip)) oip = osn
+                                    \<longrightarrow> the (dhops (rt (\<sigma> sip)) oip) \<le> hops
+                                         \<or> the (flag (rt (\<sigma> sip)) oip) = inv)"
+         and after: "\<sigma>' i = \<sigma> i\<lparr>rt := update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})\<rparr>"
+      have "\<forall>dip. dip \<in> vD (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {}))
+                \<and> dip \<in> vD (rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) oip
+                                      (osn, kno, val, Suc hops, sip, {})) dip))))
+                \<and> the (nhop (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})) dip) \<noteq> dip
+             \<longrightarrow> update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})
+                 \<sqsubset>\<^bsub>dip\<^esub>
+                 rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})) dip)))"
+      proof clarify
+        fix dip
+        assume a2: "dip\<in>vD(update (rt (\<sigma> i)) oip (osn, kno, val, Suc (hops), sip, {}))"
+           and a3: "dip\<in>vD(rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) oip
+                                                        (osn, kno, val, Suc hops, sip, {})) dip))))"
+           and a4: "the (nhop (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})) dip) \<noteq> dip"
+        from ow have a5: "\<forall>j. j \<noteq> i \<longrightarrow> \<sigma> j = \<sigma>' j" by auto
+        show "update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})
+              \<sqsubset>\<^bsub>dip\<^esub>
+              rt (\<sigma>' (the (nhop (update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {})) dip)))"
+          (is "?rt1 \<sqsubset>\<^bsub>dip\<^esub> ?rt2 dip")
+        proof (cases "?rt1 = rt (\<sigma> i)")
+          assume nochange [simp]:
+            "update (rt (\<sigma> i)) oip (osn, kno, val, Suc hops, sip, {}) = rt (\<sigma> i)"
+
+          from after have "\<sigma>' i = \<sigma> i" by simp
+          with a5 have "\<forall>j. \<sigma> j = \<sigma>' j" by metis
+
+          from a2 have "dip\<in>vD (rt (\<sigma> i))" by simp
+          moreover from a3 have "dip\<in>vD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))"
+            using nochange and \<open>\<forall>j. \<sigma> j = \<sigma>' j\<close> by clarsimp
+          moreover from a4 have "the (nhop (rt (\<sigma> i)) dip) \<noteq> dip" by simp
+          ultimately have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))"
+            using pre by simp
+
+          hence "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (the (nhop (rt (\<sigma> i)) dip)))"
+            using \<open>\<forall>j. \<sigma> j = \<sigma>' j\<close> by simp
+          thus "?thesis" by simp
+        next
+          assume change: "?rt1 \<noteq> rt (\<sigma> i)"
+          from after a2 have "dip\<in>kD(rt (\<sigma>' i))" by auto
+          show ?thesis
+          proof (cases "dip = oip")
+            assume "dip \<noteq> oip"
+
+            with a2 have "dip\<in>vD (rt (\<sigma> i))" by auto
+            moreover with a3 a5 after and \<open>dip \<noteq> oip\<close>
+              have "dip\<in>vD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))"
+                by simp metis
+            moreover from a4 and \<open>dip \<noteq> oip\<close> have "the (nhop (rt (\<sigma> i)) dip) \<noteq> dip" by simp
+            ultimately have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))"
+              using pre by simp
+
+            with after and a5 and \<open>dip \<noteq> oip\<close> show ?thesis
+              by simp (metis rt_strictly_fresher_update_other
+                             rt_strictly_fresher_irefl)
+          next
+            assume "dip = oip"
+
+            with a4 and change have "sip \<noteq> oip" by simp
+            with a6 have "oip\<in>kD(rt (\<sigma> sip))"
+                     and "osn \<le> nsqn (rt (\<sigma> sip)) oip" by auto
+
+            from a3 change \<open>dip = oip\<close> have "oip\<in>vD(rt (\<sigma>' sip))" by simp
+            hence "the (flag (rt (\<sigma>' sip)) oip) = val" by simp
+
+            from \<open>oip\<in>kD(rt (\<sigma> sip))\<close>
+            have "osn < nsqn (rt (\<sigma>' sip)) oip \<or> (osn = nsqn (rt (\<sigma>' sip)) oip
+                                                   \<and> the (dhops (rt (\<sigma>' sip)) oip) \<le> hops)"
+            proof
+              assume "oip\<in>vD(rt (\<sigma> sip))"
+              hence "the (flag (rt (\<sigma> sip)) oip) = val" by simp
+              with a6 \<open>sip \<noteq> oip\<close> have "nsqn (rt (\<sigma> sip)) oip = osn \<longrightarrow>
+                                          the (dhops (rt (\<sigma> sip)) oip) \<le> hops"
+                by simp
+              show ?thesis
+              proof (cases "sip = i")
+                assume "sip \<noteq> i"
+                with a5 have "\<sigma> sip = \<sigma>' sip" by simp
+                with \<open>osn \<le> nsqn (rt (\<sigma> sip)) oip\<close>
+                 and \<open>nsqn (rt (\<sigma> sip)) oip = osn \<longrightarrow> the (dhops (rt (\<sigma> sip)) oip) \<le> hops\<close>
+                show ?thesis by auto
+              next
+                \<comment> \<open>alternative to using @{text sip_not_ip}\<close>
+                assume [simp]: "sip = i"
+                have "?rt1 = rt (\<sigma> i)"
+                proof (rule update_cases_kD, simp_all)
+                  from \<open>Suc 0 \<le> osn\<close> show "0 < osn" by simp
+                next
+                  from \<open>oip\<in>kD(rt (\<sigma> sip))\<close> and \<open>sip = i\<close> show "oip\<in>kD(rt (\<sigma> i))"
+                    by simp
+                next
+                  assume "sqn (rt (\<sigma> i)) oip < osn"
+                  also from \<open>osn \<le> nsqn (rt (\<sigma> sip)) oip\<close>
+                    have "... \<le> nsqn (rt (\<sigma> i)) oip" by simp
+                  also have "... \<le> sqn (rt (\<sigma> i)) oip"
+                    by (rule nsqn_sqn)
+                  finally have "sqn (rt (\<sigma> i)) oip < sqn (rt (\<sigma> i)) oip" .
+                  hence False by simp
+                  thus "(\<lambda>a. if a = oip
+                             then Some (osn, kno, val, Suc hops, i, \<pi>\<^sub>7 (the (rt (\<sigma> i) oip)))
+                             else rt (\<sigma> i) a) = rt (\<sigma> i)" ..
+                next
+                  assume "sqn (rt (\<sigma> i)) oip = osn"
+                     and "Suc hops < the (dhops (rt (\<sigma> i)) oip)"
+                  from this(1) and \<open>oip \<in> vD (rt (\<sigma> sip))\<close> have "nsqn (rt (\<sigma> i)) oip = osn"
+                    by simp
+                  with \<open>nsqn (rt (\<sigma> sip)) oip = osn \<longrightarrow> the (dhops (rt (\<sigma> sip)) oip) \<le> hops\<close>
+                    have "the (dhops (rt (\<sigma> i)) oip) \<le> hops" by simp
+                  with \<open>Suc hops < the (dhops (rt (\<sigma> i)) oip)\<close> have False by simp
+                  thus "(\<lambda>a. if a = oip
+                             then Some (osn, kno, val, Suc hops, i, \<pi>\<^sub>7 (the (rt (\<sigma> i) oip)))
+                             else rt (\<sigma> i) a) = rt (\<sigma> i)" ..
+                next
+                  assume "the (flag (rt (\<sigma> i)) oip) = inv"
+                  with \<open>the (flag (rt (\<sigma> sip)) oip) = val\<close> have False by simp
+                  thus "(\<lambda>a. if a = oip
+                             then Some (osn, kno, val, Suc hops, i, \<pi>\<^sub>7 (the (rt (\<sigma> i) oip)))
+                             else rt (\<sigma> i) a) = rt (\<sigma> i)" ..
+                next
+                  from \<open>oip\<in>kD(rt (\<sigma> sip))\<close>
+                    show "(\<lambda>a. if a = oip then Some (the (rt (\<sigma> i) oip)) else rt (\<sigma> i) a) = rt (\<sigma> i)"
+                      by (auto dest!: kD_Some)
+                qed
+                with change have False ..
+                thus ?thesis ..
+              qed
+            next
+              assume "oip\<in>iD(rt (\<sigma> sip))"
+              with \<open>the (flag (rt (\<sigma>' sip)) oip) = val\<close> and a5 have "sip = i"
+                by (metis f.distinct(1) iD_flag_is_inv)
+              from \<open>oip\<in>iD(rt (\<sigma> sip))\<close> have "the (flag (rt (\<sigma> sip)) oip) = inv" by auto
+              with \<open>sip = i\<close> \<open>Suc 0 \<le> osn\<close> change after \<open>oip\<in>kD(rt (\<sigma> sip))\<close>
+                have "nsqn (rt (\<sigma> sip)) oip < nsqn (rt (\<sigma>' sip)) oip"
+                  unfolding update_def
+                  by (clarsimp split: option.split_asm if_split_asm)
+                     (auto simp: sqn_def)
+              with \<open>osn \<le> nsqn (rt (\<sigma> sip)) oip\<close> have "osn < nsqn (rt (\<sigma>' sip)) oip"
+                by simp
+              thus ?thesis ..
+            qed
+            thus ?thesis
+            proof
+              assume osnlt: "osn < nsqn (rt (\<sigma>' sip)) oip"
+              from \<open>dip\<in>kD(rt (\<sigma>' i))\<close> and \<open>dip = oip\<close> have "dip \<in> kD (?rt1)" by simp
+              moreover from a3 have "dip \<in> kD(?rt2 dip)" by simp
+              moreover have "nsqn ?rt1 dip < nsqn (?rt2 dip) dip"
+                proof -
+                  have "nsqn ?rt1 oip = osn"
+                    by (simp add: \<open>dip = oip\<close> nsqn_update_changed_kno_val [OF change [THEN not_sym]])
+                  also have "... < nsqn (rt (\<sigma>' sip)) oip" using osnlt .
+                  also have  "... = nsqn (?rt2 oip) oip" by (simp add: change)
+                  finally show ?thesis
+                    using \<open>dip = oip\<close> by simp
+                qed
+              ultimately show ?thesis
+                by (rule rt_strictly_fresher_ltI)
+            next
+              assume osneq: "osn = nsqn (rt (\<sigma>' sip)) oip \<and> the (dhops (rt (\<sigma>' sip)) oip) \<le> hops"
+
+              have "oip\<in>kD(?rt1)" by simp
+              moreover from a3 \<open>dip = oip\<close> have "oip\<in>kD(?rt2 oip)" by simp
+
+              moreover have "nsqn ?rt1 oip = nsqn (?rt2 oip) oip"
+              proof -
+                from osneq have "osn = nsqn (rt (\<sigma>' sip)) oip" ..
+                also have "osn = nsqn ?rt1 oip"
+                  by (simp add: \<open>dip = oip\<close> nsqn_update_changed_kno_val [OF change [THEN not_sym]])
+                also have  "nsqn (rt (\<sigma>' sip)) oip = nsqn (?rt2 oip) oip"
+                  by (simp add: change)
+                finally show ?thesis .
+              qed
+
+              moreover have "\<pi>\<^sub>5(the (?rt2 oip oip)) < \<pi>\<^sub>5(the (?rt1 oip))"
+              proof -
+                from osneq have "the (dhops (rt (\<sigma>' sip)) oip) \<le> hops" ..
+                moreover from \<open>oip \<in> vD (rt (\<sigma>' sip))\<close> have "oip\<in>kD(rt (\<sigma>' sip))" by auto
+                ultimately have "\<pi>\<^sub>5(the (rt (\<sigma>' sip) oip)) \<le> hops"
+                  by (auto simp add: proj5_eq_dhops)
+                also from change after have "hops < \<pi>\<^sub>5(the (rt (\<sigma>' i) oip))"
+                  by (simp add: proj5_eq_dhops) (metis dhops_update_changed lessI)
+                finally have "\<pi>\<^sub>5(the (rt (\<sigma>' sip) oip)) < \<pi>\<^sub>5(the (rt (\<sigma>' i) oip))" .
+                with change after show ?thesis by simp
+              qed
+
+              ultimately have "?rt1 \<sqsubset>\<^bsub>oip\<^esub> ?rt2 oip"
+                by (rule rt_strictly_fresher_eqI)
+              with \<open>dip = oip\<close> show ?thesis by simp
+                qed
+          qed
+       qed
+     qed
+    } note rreq_rrep_update = this
+
+    have "opaodv i \<Turnstile> (otherwith ((=)) {i} (orecvmsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m
+                                                            \<and> msg_zhops m)),
+                       other quality_increases {i} \<rightarrow>)
+            onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V
+           (\<lambda>(\<sigma>, _). \<forall>dip. dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                           \<and> the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                           \<longrightarrow> rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))"
+      proof (inv_cterms inv add: onl_oinvariant_sterms [OF aodv_wf rreq_sip [THEN weaken]]
+                                 onl_oinvariant_sterms [OF aodv_wf rrep_sip [THEN weaken]]
+                                 onl_oinvariant_sterms [OF aodv_wf rerr_sip [THEN weaken]]
+                                 onl_oinvariant_sterms [OF aodv_wf oosn_rreq [THEN weaken]]
+                                 onl_oinvariant_sterms [OF aodv_wf odsn_rrep [THEN weaken]]
+                                 onl_oinvariant_sterms [OF aodv_wf oaddpreRT_welldefined]
+                        solve: basic update_0_unk invalidate rreq_rrep_update
+                        simp add: seqlsimp)
+        fix \<sigma> \<sigma>' p l
+        assume or: "(\<sigma>, p) \<in> oreachable (opaodv i) (?S i)  (other quality_increases {i})"
+           and "other quality_increases {i} \<sigma> \<sigma>'"
+           and ll: "l \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p"
+           and pre: "\<forall>dip. dip\<in>vD (rt (\<sigma> i))
+                           \<and> dip\<in>vD(rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip))))
+                           \<and> the (nhop (rt (\<sigma> i)) dip) \<noteq> dip
+                        \<longrightarrow> rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> (the (nhop (rt (\<sigma> i)) dip)))"
+        from this(1-2)
+          have or': "(\<sigma>', p) \<in> oreachable (opaodv i) (?S i)  (other quality_increases {i})"
+            by - (rule oreachable_other')
+
+        from or and ll have next_hop: "\<forall>dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                             in dip \<in> kD(rt (\<sigma> i)) \<and> nhip \<noteq> dip
+                                             \<longrightarrow> dip \<in> kD(rt (\<sigma> nhip))
+                                                 \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> nhip)) dip"
+          by (auto dest!: onl_oinvariant_weakenD [OF seq_compare_next_hop'])
+
+        from or and ll have unk_hops_one: "\<forall>dip\<in>kD (rt (\<sigma> i)). sqn (rt (\<sigma> i)) dip = 0
+                                                \<longrightarrow> sqnf (rt (\<sigma> i)) dip = unk
+                                                    \<and> the (dhops (rt (\<sigma> i)) dip) = 1
+                                                    \<and> the (nhop (rt (\<sigma> i)) dip) = dip"
+          by (auto dest!: onl_oinvariant_weakenD [OF ozero_seq_unk_hops_one
+                                                     [OF oaodv_trans aodv_trans]]
+                          otherwith_actionD
+                          simp: seqlsimp)
+
+        from \<open>other quality_increases {i} \<sigma> \<sigma>'\<close> have "\<sigma>' i = \<sigma> i" by auto
+        hence "quality_increases (\<sigma> i) (\<sigma>' i)" by auto
+        with \<open>other quality_increases {i} \<sigma> \<sigma>'\<close> have "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+          by - (erule otherE, metis singleton_iff)
+
+        show "\<forall>dip. dip \<in> vD (rt (\<sigma>' i))
+                  \<and> dip \<in> vD (rt (\<sigma>' (the (nhop (rt (\<sigma>' i)) dip))))
+                  \<and> the (nhop (rt (\<sigma>' i)) dip) \<noteq> dip
+              \<longrightarrow> rt (\<sigma>' i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (the (nhop (rt (\<sigma>' i)) dip)))"
+        proof clarify
+          fix dip
+          assume "dip\<in>vD(rt (\<sigma>' i))"
+             and "dip\<in>vD(rt (\<sigma>' (the (nhop (rt (\<sigma>' i)) dip))))"
+             and "the (nhop (rt (\<sigma>' i)) dip) \<noteq> dip"
+          from this(1) and \<open>\<sigma>' i = \<sigma> i\<close> have "dip\<in>vD(rt (\<sigma> i))"
+                                         and "dip\<in>kD(rt (\<sigma> i))"
+            by auto
+
+          from \<open>the (nhop (rt (\<sigma>' i)) dip) \<noteq> dip\<close> and \<open>\<sigma>' i = \<sigma> i\<close>
+            have "the (nhop (rt (\<sigma> i)) dip) \<noteq> dip" (is "?nhip \<noteq> _") by simp
+          with \<open>dip\<in>kD(rt (\<sigma> i))\<close> and next_hop
+            have "dip\<in>kD(rt (\<sigma> (?nhip)))"
+             and nsqns: "nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> ?nhip)) dip"
+               by (auto simp: Let_def)
+
+          have "0 < sqn (rt (\<sigma> i)) dip"
+            proof (rule neq0_conv [THEN iffD1, OF notI])
+              assume "sqn (rt (\<sigma> i)) dip = 0"
+              with \<open>dip\<in>kD(rt (\<sigma> i))\<close> and unk_hops_one
+                have "?nhip = dip" by simp
+              with \<open>?nhip \<noteq> dip\<close> show False ..
+            qed
+          also have "... = nsqn (rt (\<sigma> i)) dip"
+            by (rule vD_nsqn_sqn [OF \<open>dip\<in>vD(rt (\<sigma> i))\<close>, THEN sym])
+          also have "... \<le> nsqn (rt (\<sigma> ?nhip)) dip"
+            by (rule nsqns)
+          also have "... \<le> sqn (rt (\<sigma> ?nhip)) dip"
+            by (rule nsqn_sqn)
+          finally have "0 < sqn (rt (\<sigma> ?nhip)) dip" .
+
+          have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' ?nhip)"
+          proof (cases "dip\<in>vD(rt (\<sigma> ?nhip))")
+            assume "dip\<in>vD(rt (\<sigma> ?nhip))"
+            with pre \<open>dip\<in>vD(rt (\<sigma> i))\<close> and \<open>?nhip \<noteq> dip\<close>
+              have "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> ?nhip)" by auto
+            moreover from \<open>\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)\<close>
+              have "quality_increases (\<sigma> ?nhip) (\<sigma>' ?nhip)" ..
+            ultimately show ?thesis
+              using \<open>dip\<in>kD(rt (\<sigma> ?nhip))\<close>
+              by (rule strictly_fresher_quality_increases_right)
+          next
+            assume "dip\<notin>vD(rt (\<sigma> ?nhip))"
+            with \<open>dip\<in>kD(rt (\<sigma> ?nhip))\<close> have "dip\<in>iD(rt (\<sigma> ?nhip))" ..
+            hence "the (flag (rt (\<sigma> ?nhip)) dip) = inv"
+              by auto
+            have "nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> ?nhip)) dip"
+              by (rule nsqns)
+            also from \<open>dip\<in>iD(rt (\<sigma> ?nhip))\<close>
+              have "... = sqn (rt (\<sigma> ?nhip)) dip - 1" ..
+            also have "... < sqn (rt (\<sigma>' ?nhip)) dip"
+              proof -
+                from \<open>\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)\<close>
+                  have "quality_increases (\<sigma> ?nhip) (\<sigma>' ?nhip)" ..
+                hence "\<forall>ip. sqn (rt (\<sigma> ?nhip)) ip \<le> sqn (rt (\<sigma>' ?nhip)) ip" by auto
+                hence "sqn (rt (\<sigma> ?nhip)) dip \<le> sqn (rt (\<sigma>' ?nhip)) dip" ..
+                with \<open>0 < sqn (rt (\<sigma> ?nhip)) dip\<close> show ?thesis by auto
+              qed
+            also have "... = nsqn (rt (\<sigma>' ?nhip)) dip"
+              proof (rule vD_nsqn_sqn [THEN sym])
+                from \<open>dip\<in>vD(rt (\<sigma>' (the (nhop (rt (\<sigma>' i)) dip))))\<close> and \<open>\<sigma>' i = \<sigma> i\<close>
+                  show "dip\<in>vD(rt (\<sigma>' ?nhip))" by simp
+              qed
+            finally have "nsqn (rt (\<sigma> i)) dip < nsqn (rt (\<sigma>' ?nhip)) dip" .
+
+            moreover from \<open>dip\<in>vD(rt (\<sigma>' (the (nhop (rt (\<sigma>' i)) dip))))\<close> and \<open>\<sigma>' i = \<sigma> i\<close>
+              have "dip\<in>kD(rt (\<sigma>' ?nhip))" by auto
+            ultimately show "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' ?nhip)"
+              using \<open>dip\<in>kD(rt (\<sigma> i))\<close> by - (rule rt_strictly_fresher_ltI)
+          qed
+          with \<open>\<sigma>' i = \<sigma> i\<close> show "rt (\<sigma>' i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' (the (nhop (rt (\<sigma>' i)) dip)))"
+            by simp
+        qed
+      qed
+    thus ?thesis unfolding Let_def .
+  qed
+
+lemma seq_compare_next_hop:
+  fixes w
+  shows "opaodv i \<Turnstile> (otherwith ((=)) {i} (orecvmsg msg_fresh),
+                      other quality_increases {i} \<rightarrow>)
+                       global (\<lambda>\<sigma>. \<forall>dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                         in dip \<in> kD(rt (\<sigma> i)) \<and> nhip \<noteq> dip \<longrightarrow>
+                                            dip \<in> kD(rt (\<sigma> nhip))
+                                            \<and> nsqn (rt (\<sigma> i)) dip \<le> nsqn (rt (\<sigma> nhip)) dip)"
+  by (rule oinvariant_weakenE [OF seq_compare_next_hop']) (auto dest!: onlD)
+
+lemma seq_nhop_quality_increases:
+  shows "opaodv i \<Turnstile> (otherwith ((=)) {i}
+                        (orecvmsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)),
+                         other quality_increases {i} \<rightarrow>)
+                        global (\<lambda>\<sigma>. \<forall>dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                          in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                                             \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+  by (rule oinvariant_weakenE [OF seq_nhop_quality_increases']) (auto dest!: onlD)
+
+end

formal/afp/AODV/variants/a_norreqid/A_Loop_Freedom.thy

@@ -0,0 +1,123 @@
+(*  Title:       variants/a_norreqid/Loop_Freedom.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "Routing graphs and loop freedom"
+
+theory A_Loop_Freedom
+imports A_Aodv_Predicates A_Fresher
+begin
+
+text \<open>Define the central theorem that relates an invariant over network states to the absence
+      of loops in the associate routing graph.\<close>
+
+definition
+  rt_graph :: "(ip \<Rightarrow> state) \<Rightarrow> ip \<Rightarrow> ip rel"
+where
+  "rt_graph \<sigma> = (\<lambda>dip.
+     {(ip, ip') | ip ip' dsn dsk hops pre.
+        ip \<noteq> dip \<and> rt (\<sigma> ip) dip = Some (dsn, dsk, val, hops, ip', pre)})"
+
+text \<open>Given the state of a network @{term \<sigma>}, a routing graph for a given destination
+      ip address @{term dip} abstracts the details of routing tables into nodes
+      (ip addresses) and vertices (valid routes between ip addresses).\<close>
+
+lemma rt_graphE [elim]:
+    fixes n dip ip ip'
+  assumes "(ip, ip') \<in> rt_graph \<sigma> dip"
+    shows "ip \<noteq> dip \<and> (\<exists>r. rt (\<sigma> ip) = r
+                            \<and> (\<exists>dsn dsk hops pre. r dip = Some (dsn, dsk, val, hops, ip', pre)))"
+  using assms unfolding rt_graph_def by auto
+
+lemma rt_graph_vD [dest]:
+  "\<And>ip ip' \<sigma> dip. (ip, ip') \<in> rt_graph \<sigma> dip \<Longrightarrow> dip \<in> vD(rt (\<sigma> ip))"
+  unfolding rt_graph_def vD_def by auto
+
+lemma rt_graph_vD_trans [dest]:
+  "\<And>ip ip' \<sigma> dip. (ip, ip') \<in> (rt_graph \<sigma> dip)\<^sup>+ \<Longrightarrow> dip \<in> vD(rt (\<sigma> ip))"
+  by (erule converse_tranclE) auto
+
+lemma rt_graph_not_dip [dest]:
+  "\<And>ip ip' \<sigma> dip. (ip, ip') \<in> rt_graph \<sigma> dip \<Longrightarrow> ip \<noteq> dip"
+  unfolding rt_graph_def by auto
+
+lemma rt_graph_not_dip_trans [dest]:
+  "\<And>ip ip' \<sigma> dip. (ip, ip') \<in> (rt_graph \<sigma> dip)\<^sup>+ \<Longrightarrow> ip \<noteq> dip"
+  by (erule converse_tranclE) auto
+
+text "NB: the property below cannot be lifted to the transitive closure"
+
+lemma rt_graph_nhip_is_nhop [dest]:
+  "\<And>ip ip' \<sigma> dip. (ip, ip') \<in> rt_graph \<sigma> dip \<Longrightarrow> ip' = the (nhop (rt (\<sigma> ip)) dip)"
+  unfolding rt_graph_def by auto
+
+theorem inv_to_loop_freedom:
+  assumes "\<forall>i dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                   in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                      \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip))"
+    shows "\<forall>dip. irrefl ((rt_graph \<sigma> dip)\<^sup>+)"
+  using assms proof (intro allI)
+    fix \<sigma> :: "ip \<Rightarrow> state" and dip
+    assume inv: "\<forall>ip dip.
+                  let nhip = the (nhop (rt (\<sigma> ip)) dip)
+                  in dip \<in> vD(rt (\<sigma> ip)) \<inter> vD(rt (\<sigma> nhip)) \<and>
+                     nhip \<noteq> dip \<longrightarrow> rt (\<sigma> ip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip)"
+    { fix ip ip'
+      assume "(ip, ip') \<in> (rt_graph \<sigma> dip)\<^sup>+"
+         and "dip \<in> vD(rt (\<sigma> ip'))"
+         and "ip' \<noteq> dip"
+       hence "rt (\<sigma> ip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> ip')"
+         proof induction
+           fix nhip
+           assume "(ip, nhip) \<in> rt_graph \<sigma> dip"
+              and "dip \<in> vD(rt (\<sigma> nhip))"
+              and "nhip \<noteq> dip"
+           from \<open>(ip, nhip) \<in> rt_graph \<sigma> dip\<close> have "dip \<in> vD(rt (\<sigma> ip))"
+                                               and "nhip = the (nhop (rt (\<sigma> ip)) dip)"
+             by auto
+           from \<open>dip \<in> vD(rt (\<sigma> ip))\<close> and \<open>dip \<in> vD(rt (\<sigma> nhip))\<close>
+             have "dip \<in> vD(rt (\<sigma> ip)) \<inter> vD(rt (\<sigma> nhip))" ..
+           with \<open>nhip = the (nhop (rt (\<sigma> ip)) dip)\<close>
+                and \<open>nhip \<noteq> dip\<close>
+                and inv
+             show "rt (\<sigma> ip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip)"
+             by (clarsimp simp: Let_def)
+         next
+           fix nhip nhip'
+           assume "(ip, nhip) \<in> (rt_graph \<sigma> dip)\<^sup>+"
+              and "(nhip, nhip') \<in> rt_graph \<sigma> dip"
+              and IH: "\<lbrakk> dip \<in> vD(rt (\<sigma> nhip)); nhip \<noteq> dip \<rbrakk> \<Longrightarrow> rt (\<sigma> ip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip)"
+              and "dip \<in> vD(rt (\<sigma> nhip'))"
+              and "nhip' \<noteq> dip"
+           from \<open>(nhip, nhip') \<in> rt_graph \<sigma> dip\<close> have 1: "dip \<in> vD(rt (\<sigma> nhip))"
+                                                  and 2: "nhip \<noteq> dip"
+                                                  and "nhip' = the (nhop (rt (\<sigma> nhip)) dip)"
+             by auto
+           from 1 2 have "rt (\<sigma> ip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip)" by (rule IH)
+           also have "rt (\<sigma> nhip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip')"
+             proof -
+               from \<open>dip \<in> vD(rt (\<sigma> nhip))\<close> and \<open>dip \<in> vD(rt (\<sigma> nhip'))\<close>
+                 have "dip \<in> vD(rt (\<sigma> nhip)) \<inter> vD(rt (\<sigma> nhip'))" ..
+               with \<open>nhip' \<noteq> dip\<close>
+                    and \<open>nhip' = the (nhop (rt (\<sigma> nhip)) dip)\<close>
+                    and inv
+                 show "rt (\<sigma> nhip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip')"
+                   by (clarsimp simp: Let_def)
+             qed
+           finally show "rt (\<sigma> ip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip')" .
+         qed } note fresher = this
+
+    show "irrefl ((rt_graph \<sigma> dip)\<^sup>+)"
+    unfolding irrefl_def proof (intro allI notI)
+      fix ip
+      assume "(ip, ip) \<in> (rt_graph \<sigma> dip)\<^sup>+"
+      moreover then have "dip \<in> vD(rt (\<sigma> ip))"
+                     and "ip \<noteq> dip"
+        by auto
+      ultimately have "rt (\<sigma> ip) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> ip)" by (rule fresher)
+      thus False by simp
+    qed
+  qed
+
+end

formal/afp/AODV/variants/a_norreqid/A_Norreqid.thy

@@ -0,0 +1,25 @@
+(*  Title:       variants/a_norreqid/A_Norreqid.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+    Author:      Peter Höfner, NICTA
+*)
+
+theory %invisible A_Norreqid
+imports "../../Aodv_Basic"
+begin
+
+chapter "Variant A: Skipping the RREQ ID"
+
+text \<open>
+  Explanation~\cite[\textsection 10.1]{FehnkerEtAl:AWN:2013}:
+  AODV does not need the route request identifier. This number, in 
+  combination with the IP address of the originator, is used to identify 
+  every RREQ message in a unique way. This variant shows that the 
+  combination of the originator's IP address and its sequence number is just 
+  as suited to uniquely determine the route request to which the message 
+  belongs. Hence, the route request identifier field is not required. This 
+  can then reduce the size of the RREQ message.
+\<close>
+
+end %invisible
+

formal/afp/AODV/variants/a_norreqid/A_OAodv.thy

@@ -0,0 +1,47 @@
+(*  Title:       variants/a_norreqid/OAodv.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "The `open' AODV model"
+
+theory A_OAodv
+imports A_Aodv AWN.OAWN_SOS_Labels AWN.OAWN_Convert
+begin
+
+text \<open>Definitions for stating and proving global network properties over individual processes.\<close>
+
+definition \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V' :: "((ip \<Rightarrow> state) \<times> ((state, msg, pseqp, pseqp label) seqp)) set"
+where "\<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V' \<equiv> {(\<lambda>i. aodv_init i, \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PAodv)}"
+
+abbreviation opaodv
+  :: "ip \<Rightarrow> ((ip \<Rightarrow> state) \<times> (state, msg, pseqp, pseqp label) seqp, msg seq_action) automaton"
+where
+  "opaodv i \<equiv> \<lparr> init = \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V', trans = oseqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i \<rparr>"
+
+lemma initiali_aodv [intro!, simp]: "initiali i (init (opaodv i)) (init (paodv i))"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'_def by rule simp_all
+
+lemma oaodv_control_within [simp]: "control_within \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (init (opaodv i))"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'_def by (rule control_withinI) (auto simp del: \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_simps)
+
+lemma \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'_labels [simp]: "(\<sigma>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V' \<Longrightarrow>  labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p = {PAodv-:0}"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'_def by simp
+
+lemma oaodv_init_kD_empty [simp]:
+  "(\<sigma>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V' \<Longrightarrow> kD (rt (\<sigma> i)) = {}"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'_def kD_def by simp
+
+lemma oaodv_init_vD_empty [simp]:
+  "(\<sigma>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V' \<Longrightarrow> vD (rt (\<sigma> i)) = {}"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'_def vD_def by simp
+
+lemma oaodv_trans: "trans (opaodv i) = oseqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i"
+  by simp
+
+declare
+  oseq_invariant_ctermsI [OF aodv_wf oaodv_control_within aodv_simple_labels oaodv_trans, cterms_intros]
+  oseq_step_invariant_ctermsI [OF aodv_wf oaodv_control_within aodv_simple_labels oaodv_trans, cterms_intros]
+
+end
+

formal/afp/AODV/variants/a_norreqid/A_Quality_Increases.thy

@@ -0,0 +1,457 @@
+(*  Title:       variants/a_norreqid/Quality_Increases.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+    Author:      Peter Höfner, NICTA
+*)
+
+section "The quality increases predicate"
+
+theory A_Quality_Increases
+imports A_Aodv_Predicates A_Fresher
+begin
+
+definition quality_increases :: "state \<Rightarrow> state \<Rightarrow> bool"
+where "quality_increases \<xi> \<xi>' \<equiv> (\<forall>dip\<in>kD(rt \<xi>). dip \<in> kD(rt \<xi>') \<and> rt \<xi> \<sqsubseteq>\<^bsub>dip\<^esub> rt \<xi>')
+                                               \<and> (\<forall>dip. sqn (rt \<xi>) dip \<le> sqn (rt \<xi>') dip)"
+
+lemma quality_increasesI [intro!]:
+  assumes "\<And>dip. dip \<in> kD(rt \<xi>) \<Longrightarrow> dip \<in> kD(rt \<xi>')"
+      and "\<And>dip. \<lbrakk> dip \<in> kD(rt \<xi>); dip \<in> kD(rt \<xi>') \<rbrakk> \<Longrightarrow> rt \<xi> \<sqsubseteq>\<^bsub>dip\<^esub> rt \<xi>'"          
+      and "\<And>dip. sqn (rt \<xi>) dip \<le> sqn (rt \<xi>') dip"
+    shows "quality_increases \<xi> \<xi>'"
+  unfolding quality_increases_def using assms by clarsimp
+
+lemma quality_increasesE [elim]:
+    fixes dip
+  assumes "quality_increases \<xi> \<xi>'"
+      and "dip\<in>kD(rt \<xi>)"
+      and "\<lbrakk> dip \<in> kD(rt \<xi>'); rt \<xi> \<sqsubseteq>\<^bsub>dip\<^esub> rt \<xi>'; sqn (rt \<xi>) dip \<le> sqn (rt \<xi>') dip \<rbrakk> \<Longrightarrow> R dip \<xi> \<xi>'"
+    shows "R dip \<xi> \<xi>'"
+  using assms unfolding quality_increases_def by clarsimp
+
+lemma quality_increases_rt_fresherD [dest]:
+    fixes ip
+  assumes "quality_increases \<xi> \<xi>'"
+      and "ip\<in>kD(rt \<xi>)"
+    shows "rt \<xi> \<sqsubseteq>\<^bsub>ip\<^esub> rt \<xi>'"
+  using assms by auto
+
+lemma quality_increases_sqnE [elim]:
+    fixes dip
+  assumes "quality_increases \<xi> \<xi>'"
+      and "sqn (rt \<xi>) dip \<le> sqn (rt \<xi>') dip \<Longrightarrow> R dip \<xi> \<xi>'"
+    shows "R dip \<xi> \<xi>'"
+  using assms unfolding quality_increases_def by clarsimp
+
+lemma quality_increases_refl [intro, simp]: "quality_increases \<xi> \<xi>"
+  by rule simp_all
+
+lemma strictly_fresher_quality_increases_right [elim]:
+    fixes \<sigma> \<sigma>' dip
+  assumes "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip)"                       
+      and qinc: "quality_increases (\<sigma> nhip) (\<sigma>' nhip)"
+      and "dip\<in>kD(rt (\<sigma> nhip))"
+    shows "rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma>' nhip)"
+  proof -
+    from qinc have "rt (\<sigma> nhip) \<sqsubseteq>\<^bsub>dip\<^esub> rt (\<sigma>' nhip)" using \<open>dip\<in>kD(rt (\<sigma> nhip))\<close>
+      by auto
+    with \<open>rt (\<sigma> i) \<sqsubset>\<^bsub>dip\<^esub> rt (\<sigma> nhip)\<close> show ?thesis ..
+  qed
+
+lemma kD_quality_increases [elim]:
+  assumes "i\<in>kD(rt \<xi>)"
+      and "quality_increases \<xi> \<xi>'"
+    shows "i\<in>kD(rt \<xi>')"
+  using assms by auto
+
+lemma kD_nsqn_quality_increases [elim]:
+  assumes "i\<in>kD(rt \<xi>)"
+      and "quality_increases \<xi> \<xi>'"
+    shows "i\<in>kD(rt \<xi>') \<and> nsqn (rt \<xi>) i \<le> nsqn (rt \<xi>') i"
+  proof -
+    from assms have "i\<in>kD(rt \<xi>')" ..
+    moreover with assms have "rt \<xi> \<sqsubseteq>\<^bsub>i\<^esub> rt \<xi>'" by auto
+    ultimately have "nsqn (rt \<xi>) i \<le> nsqn (rt \<xi>') i"
+      using \<open>i\<in>kD(rt \<xi>)\<close> by - (erule(2) rt_fresher_imp_nsqn_le)
+    with \<open>i\<in>kD(rt \<xi>')\<close> show ?thesis ..
+  qed
+
+lemma nsqn_quality_increases [elim]:
+  assumes "i\<in>kD(rt \<xi>)"
+      and "quality_increases \<xi> \<xi>'"
+    shows "nsqn (rt \<xi>) i \<le> nsqn (rt \<xi>') i"
+  using assms by (rule kD_nsqn_quality_increases [THEN conjunct2])
+
+lemma kD_nsqn_quality_increases_trans [elim]:
+  assumes "i\<in>kD(rt \<xi>)"
+      and "s \<le> nsqn (rt \<xi>) i"
+      and "quality_increases \<xi> \<xi>'"
+    shows "i\<in>kD(rt \<xi>') \<and> s \<le> nsqn (rt \<xi>') i"
+  proof
+    from \<open>i\<in>kD(rt \<xi>)\<close> and \<open>quality_increases \<xi> \<xi>'\<close> show "i\<in>kD(rt \<xi>')" ..
+  next
+    from \<open>i\<in>kD(rt \<xi>)\<close> and \<open>quality_increases \<xi> \<xi>'\<close> have "nsqn (rt \<xi>) i \<le> nsqn (rt \<xi>') i" ..
+    with \<open>s \<le> nsqn (rt \<xi>) i\<close> show "s \<le> nsqn (rt \<xi>') i" by (rule le_trans)
+  qed
+
+lemma nsqn_quality_increases_nsqn_lt_lt [elim]:
+  assumes "i\<in>kD(rt \<xi>)"
+      and "quality_increases \<xi> \<xi>'"
+     and "s < nsqn (rt \<xi>) i"
+    shows "s < nsqn (rt \<xi>') i"
+  proof -
+    from assms(1-2) have "nsqn (rt \<xi>) i \<le> nsqn (rt \<xi>') i" ..
+    with \<open>s < nsqn (rt \<xi>) i\<close> show "s < nsqn (rt \<xi>') i" by simp
+  qed
+
+lemma nsqn_quality_increases_dhops [elim]:
+  assumes "i\<in>kD(rt \<xi>)"
+      and "quality_increases \<xi> \<xi>'"
+      and "nsqn (rt \<xi>) i = nsqn (rt \<xi>') i"
+    shows "the (dhops (rt \<xi>) i) \<ge> the (dhops (rt \<xi>') i)"
+  using assms unfolding quality_increases_def
+  by (clarsimp) (drule(1) bspec, clarsimp simp: rt_fresher_def2)
+
+lemma nsqn_quality_increases_nsqn_eq_le [elim]:
+  assumes "i\<in>kD(rt \<xi>)"
+      and "quality_increases \<xi> \<xi>'"
+      and "s = nsqn (rt \<xi>) i"
+    shows "s < nsqn (rt \<xi>') i \<or> (s = nsqn (rt \<xi>') i \<and> the (dhops (rt \<xi>) i) \<ge> the (dhops (rt \<xi>') i))"
+  using assms by (metis nat_less_le nsqn_quality_increases nsqn_quality_increases_dhops)
+
+lemma quality_increases_rreq_rrep_props [elim]:
+    fixes sn ip hops sip
+  assumes qinc: "quality_increases (\<sigma> sip) (\<sigma>' sip)"
+      and "1 \<le> sn"
+      and *: "ip\<in>kD(rt (\<sigma> sip)) \<and> sn \<le> nsqn (rt (\<sigma> sip)) ip
+                                \<and> (nsqn (rt (\<sigma> sip)) ip = sn
+                                    \<longrightarrow> (the (dhops (rt (\<sigma> sip)) ip) \<le> hops
+                                          \<or> the (flag (rt (\<sigma> sip)) ip) = inv))"
+    shows "ip\<in>kD(rt (\<sigma>' sip)) \<and> sn \<le> nsqn (rt (\<sigma>' sip)) ip
+                              \<and> (nsqn (rt (\<sigma>' sip)) ip = sn
+                                  \<longrightarrow> (the (dhops (rt (\<sigma>' sip)) ip) \<le> hops
+                                        \<or> the (flag (rt (\<sigma>' sip)) ip) = inv))"
+      (is "_ \<and> ?nsqnafter")
+  proof -
+    from *  obtain "ip\<in>kD(rt (\<sigma> sip))" and "sn \<le> nsqn (rt (\<sigma> sip)) ip" by auto
+
+    from \<open>quality_increases (\<sigma> sip) (\<sigma>' sip)\<close>
+       have "sqn (rt (\<sigma> sip)) ip \<le> sqn (rt (\<sigma>' sip)) ip" ..
+    from \<open>quality_increases (\<sigma> sip) (\<sigma>' sip)\<close> and \<open>ip\<in>kD (rt (\<sigma> sip))\<close>
+      have "ip\<in>kD (rt (\<sigma>' sip))" ..
+
+    from \<open>sn \<le> nsqn (rt (\<sigma> sip)) ip\<close> have ?nsqnafter
+    proof
+      assume "sn < nsqn (rt (\<sigma> sip)) ip"
+      also from \<open>ip\<in>kD(rt (\<sigma> sip))\<close> and \<open>quality_increases (\<sigma> sip) (\<sigma>' sip)\<close>
+        have "... \<le> nsqn (rt (\<sigma>' sip)) ip" ..
+      finally have "sn < nsqn (rt (\<sigma>' sip)) ip" .
+      thus ?thesis by simp
+    next
+      assume "sn = nsqn (rt (\<sigma> sip)) ip"
+      with \<open>ip\<in>kD(rt (\<sigma> sip))\<close> and \<open>quality_increases (\<sigma> sip) (\<sigma>' sip)\<close>
+        have "sn < nsqn (rt (\<sigma>' sip)) ip
+              \<or> (sn = nsqn (rt (\<sigma>' sip)) ip
+                 \<and> the (dhops (rt (\<sigma>' sip)) ip) \<le> the (dhops (rt (\<sigma> sip)) ip))" ..
+      hence "sn < nsqn (rt (\<sigma>' sip)) ip
+              \<or> (nsqn (rt (\<sigma>' sip)) ip = sn \<and> (the (dhops (rt (\<sigma>' sip)) ip) \<le> hops
+                                                 \<or> the (flag (rt (\<sigma>' sip)) ip) = inv))"
+      proof
+        assume "sn < nsqn (rt (\<sigma>' sip)) ip" thus ?thesis ..
+      next
+        assume "sn = nsqn (rt (\<sigma>' sip)) ip
+                \<and> the (dhops (rt (\<sigma> sip)) ip) \<ge> the (dhops (rt (\<sigma>' sip)) ip)"
+        hence "sn = nsqn (rt (\<sigma>' sip)) ip"
+          and "the (dhops (rt (\<sigma>' sip)) ip) \<le> the (dhops (rt (\<sigma> sip)) ip)" by auto
+
+        from * and \<open>sn = nsqn (rt (\<sigma> sip)) ip\<close> have "the (dhops (rt (\<sigma> sip)) ip) \<le> hops
+                                                       \<or> the (flag (rt (\<sigma> sip)) ip) = inv"
+          by simp
+        thus ?thesis
+        proof
+          assume "the (dhops (rt (\<sigma> sip)) ip) \<le> hops"
+          with  \<open>the (dhops (rt (\<sigma>' sip)) ip) \<le> the (dhops (rt (\<sigma> sip)) ip)\<close>
+           have "the (dhops (rt (\<sigma>' sip)) ip) \<le> hops" by simp
+          with \<open>sn = nsqn (rt (\<sigma>' sip)) ip\<close> show ?thesis by simp
+        next
+          assume "the (flag (rt (\<sigma> sip)) ip) = inv"
+          with \<open>ip\<in>kD(rt (\<sigma> sip))\<close> have "nsqn (rt (\<sigma> sip)) ip = sqn (rt (\<sigma> sip)) ip - 1" ..
+
+          with \<open>sn \<ge> 1\<close> and \<open>sn = nsqn (rt (\<sigma> sip)) ip\<close>
+            have "sqn (rt (\<sigma> sip)) ip > 1" by simp
+
+          from \<open>ip\<in>kD(rt (\<sigma>' sip))\<close> show ?thesis
+          proof (rule vD_or_iD)
+            assume "ip\<in>iD(rt (\<sigma>' sip))"
+            hence "the (flag (rt (\<sigma>' sip)) ip) = inv" ..
+            with \<open>sn = nsqn (rt (\<sigma>' sip)) ip\<close> show ?thesis
+              by simp
+          next
+            (* the tricky case: sn = nsqn (rt (\<sigma>' sip)) ip
+                                \<and> ip\<in>iD(rt (\<sigma> sip))
+                                \<and> ip\<in>vD(rt (\<sigma>' sip)) *)
+            assume "ip\<in>vD(rt (\<sigma>' sip))"
+            hence "nsqn (rt (\<sigma>' sip)) ip = sqn (rt (\<sigma>' sip)) ip" ..
+            with \<open>sqn (rt (\<sigma> sip)) ip \<le> sqn (rt (\<sigma>' sip)) ip\<close>
+              have "nsqn (rt (\<sigma>' sip)) ip \<ge> sqn (rt (\<sigma> sip)) ip" by simp
+
+            with \<open>sqn (rt (\<sigma> sip)) ip > 1\<close>
+              have "nsqn (rt (\<sigma>' sip)) ip > sqn (rt (\<sigma> sip)) ip - 1" by simp
+            with \<open>nsqn (rt (\<sigma> sip)) ip = sqn (rt (\<sigma> sip)) ip - 1\<close>
+              have "nsqn (rt (\<sigma>' sip)) ip > nsqn (rt (\<sigma> sip)) ip" by simp
+            with \<open>sn = nsqn (rt (\<sigma> sip)) ip\<close> have "nsqn (rt (\<sigma>' sip)) ip > sn"
+              by simp
+            thus ?thesis ..
+          qed
+        qed
+      qed
+      thus ?thesis by (metis (mono_tags) le_cases not_le)
+    qed
+    with \<open>ip\<in>kD (rt (\<sigma>' sip))\<close> show "ip\<in>kD (rt (\<sigma>' sip)) \<and> ?nsqnafter" ..
+  qed
+
+lemma quality_increases_rreq_rrep_props':
+    fixes sn ip hops sip
+  assumes "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+      and "1 \<le> sn"
+      and *: "ip\<in>kD(rt (\<sigma> sip)) \<and> sn \<le> nsqn (rt (\<sigma> sip)) ip
+                                \<and> (nsqn (rt (\<sigma> sip)) ip = sn
+                                    \<longrightarrow> (the (dhops (rt (\<sigma> sip)) ip) \<le> hops
+                                          \<or> the (flag (rt (\<sigma> sip)) ip) = inv))"
+    shows "ip\<in>kD(rt (\<sigma>' sip)) \<and> sn \<le> nsqn (rt (\<sigma>' sip)) ip
+                              \<and> (nsqn (rt (\<sigma>' sip)) ip = sn
+                                  \<longrightarrow> (the (dhops (rt (\<sigma>' sip)) ip) \<le> hops
+                                        \<or> the (flag (rt (\<sigma>' sip)) ip) = inv))"
+  proof -
+    from assms(1) have "quality_increases (\<sigma> sip) (\<sigma>' sip)" ..
+    thus ?thesis using assms(2-3) by (rule quality_increases_rreq_rrep_props)
+  qed
+
+lemma rteq_quality_increases:
+  assumes "\<forall>j. j \<noteq> i \<longrightarrow> quality_increases (\<sigma> j) (\<sigma>' j)"
+      and "rt (\<sigma>' i) = rt (\<sigma> i)"
+    shows "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+  using assms by clarsimp (metis order_refl quality_increasesI rt_fresher_refl)
+
+definition msg_fresh :: "(ip \<Rightarrow> state) \<Rightarrow> msg \<Rightarrow> bool"
+where "msg_fresh \<sigma> m \<equiv>
+         case m of Rreq hopsc  _ _ _ oipc osnc sipc \<Rightarrow> osnc \<ge> 1 \<and> (sipc \<noteq> oipc \<longrightarrow>
+                       oipc\<in>kD(rt (\<sigma> sipc)) \<and> nsqn (rt (\<sigma> sipc)) oipc \<ge> osnc
+                       \<and> (nsqn (rt (\<sigma> sipc)) oipc = osnc
+                             \<longrightarrow> (hopsc \<ge> the (dhops (rt (\<sigma> sipc)) oipc)
+                                  \<or> the (flag (rt (\<sigma> sipc)) oipc) = inv)))
+                   | Rrep hopsc dipc dsnc _ sipc \<Rightarrow> dsnc \<ge> 1 \<and> (sipc \<noteq> dipc \<longrightarrow>
+                       dipc\<in>kD(rt (\<sigma> sipc)) \<and> nsqn (rt (\<sigma> sipc)) dipc \<ge> dsnc
+                       \<and> (nsqn (rt (\<sigma> sipc)) dipc = dsnc
+                             \<longrightarrow> (hopsc \<ge> the (dhops (rt (\<sigma> sipc)) dipc)
+                                   \<or> the (flag (rt (\<sigma> sipc)) dipc) = inv)))
+                   | Rerr destsc sipc \<Rightarrow> (\<forall>ripc\<in>dom(destsc). (ripc\<in>kD(rt (\<sigma> sipc))
+                                         \<and> the (destsc ripc) - 1 \<le> nsqn (rt (\<sigma> sipc)) ripc))
+                   | _ \<Rightarrow> True"
+
+lemma msg_fresh [simp]:
+  "\<And>hops dip dsn dsk oip osn sip.
+           msg_fresh \<sigma> (Rreq hops dip dsn dsk oip osn sip) =
+                            (osn \<ge> 1 \<and> (sip \<noteq> oip \<longrightarrow> oip\<in>kD(rt (\<sigma> sip))
+                                     \<and> nsqn (rt (\<sigma> sip)) oip \<ge> osn
+                                     \<and> (nsqn (rt (\<sigma> sip)) oip = osn
+                                           \<longrightarrow> (hops \<ge> the (dhops (rt (\<sigma> sip)) oip)
+                                                \<or> the (flag (rt (\<sigma> sip)) oip) = inv))))"
+  "\<And>hops dip dsn oip sip. msg_fresh \<sigma> (Rrep hops dip dsn oip sip) =
+                            (dsn \<ge> 1 \<and> (sip \<noteq> dip \<longrightarrow> dip\<in>kD(rt (\<sigma> sip))
+                                     \<and> nsqn (rt (\<sigma> sip)) dip \<ge> dsn
+                                     \<and> (nsqn (rt (\<sigma> sip)) dip = dsn
+                                           \<longrightarrow> (hops \<ge> the (dhops (rt (\<sigma> sip)) dip))
+                                                 \<or> the (flag (rt (\<sigma> sip)) dip) = inv)))"
+  "\<And>dests sip.            msg_fresh \<sigma> (Rerr dests sip) =
+                            (\<forall>ripc\<in>dom(dests). (ripc\<in>kD(rt (\<sigma> sip))
+                                     \<and> the (dests ripc) - 1 \<le> nsqn (rt (\<sigma> sip)) ripc))"
+  "\<And>d dip.                msg_fresh \<sigma> (Newpkt d dip)   = True"
+  "\<And>d dip sip.            msg_fresh \<sigma> (Pkt d dip sip)  = True"
+  unfolding msg_fresh_def by simp_all
+
+lemma msg_fresh_inc_sn [simp, elim]:
+  "msg_fresh \<sigma> m \<Longrightarrow> rreq_rrep_sn m"
+  by (cases m) simp_all
+
+lemma recv_msg_fresh_inc_sn [simp, elim]:
+  "orecvmsg (msg_fresh) \<sigma> m \<Longrightarrow> recvmsg rreq_rrep_sn m"
+  by (cases m) simp_all
+
+lemma rreq_nsqn_is_fresh [simp]:
+  fixes \<sigma> msg hops dip dsn dsk oip osn sip
+  assumes "rreq_rrep_fresh (rt (\<sigma> sip)) (Rreq hops dip dsn dsk oip osn sip)"
+      and "rreq_rrep_sn (Rreq hops  dip dsn dsk oip osn sip)"
+  shows "msg_fresh \<sigma> (Rreq hops  dip dsn dsk oip osn sip)"
+        (is "msg_fresh \<sigma> ?msg")
+  proof -
+    let ?rt = "rt (\<sigma> sip)"
+    from assms(2) have "1 \<le> osn" by simp
+    thus ?thesis
+    unfolding msg_fresh_def
+    proof (simp only: msg.case, intro conjI impI)
+      assume "sip \<noteq> oip"
+      with assms(1) show "oip \<in> kD(?rt)" by simp
+    next
+      assume "sip \<noteq> oip"
+         and "nsqn ?rt oip = osn"
+      show "the (dhops ?rt oip) \<le> hops \<or> the (flag ?rt oip) = inv"
+      proof (cases "oip\<in>vD(?rt)")
+        assume "oip\<in>vD(?rt)"
+        hence "nsqn ?rt oip = sqn ?rt oip" ..
+        with \<open>nsqn ?rt oip = osn\<close> have "sqn ?rt oip = osn" by simp
+        with assms(1) and \<open>sip \<noteq> oip\<close> have "the (dhops ?rt oip) \<le> hops"
+          by simp
+        thus ?thesis ..
+      next
+        assume "oip\<notin>vD(?rt)"
+        moreover from assms(1) and \<open>sip \<noteq> oip\<close> have "oip\<in>kD(?rt)" by simp
+        ultimately have "oip\<in>iD(?rt)" by auto
+        hence "the (flag ?rt oip) = inv" ..
+        thus ?thesis ..
+      qed
+    next
+      assume "sip \<noteq> oip"
+      with assms(1) have "osn \<le> sqn ?rt oip" by auto
+      thus "osn \<le> nsqn (rt (\<sigma> sip)) oip"
+      proof (rule nat_le_eq_or_lt)
+        assume "osn < sqn ?rt oip"
+        hence "osn \<le> sqn ?rt oip - 1" by simp
+        also have "... \<le> nsqn ?rt oip" by (rule sqn_nsqn)
+        finally show "osn \<le> nsqn ?rt oip" .
+      next
+        assume "osn = sqn ?rt oip"
+        with assms(1) and \<open>sip \<noteq> oip\<close> have "oip\<in>kD(?rt)"
+                                       and "the (flag ?rt oip) = val"
+          by auto
+        hence "nsqn ?rt oip = sqn ?rt oip" ..
+        with \<open>osn = sqn ?rt oip\<close> have "nsqn ?rt oip = osn" by simp
+        thus "osn \<le> nsqn ?rt oip" by simp
+      qed
+    qed simp
+  qed
+
+lemma rrep_nsqn_is_fresh [simp]:
+  fixes \<sigma> msg hops dip dsn oip sip
+  assumes "rreq_rrep_fresh (rt (\<sigma> sip)) (Rrep hops dip dsn oip sip)"
+      and "rreq_rrep_sn (Rrep hops dip dsn oip sip)"
+  shows "msg_fresh \<sigma> (Rrep hops dip dsn oip sip)"
+        (is "msg_fresh \<sigma> ?msg")
+  proof -
+    let ?rt = "rt (\<sigma> sip)"
+    from assms have "sip \<noteq> dip \<longrightarrow> dip\<in>kD(?rt) \<and> sqn ?rt dip = dsn \<and> the (flag ?rt dip) = val"
+      by simp
+    hence "sip \<noteq> dip \<longrightarrow> dip\<in>kD(?rt) \<and> nsqn ?rt dip \<ge> dsn"
+    by clarsimp
+    with assms show "msg_fresh \<sigma> ?msg"
+      by clarsimp
+  qed
+
+lemma rerr_nsqn_is_fresh [simp]:
+  fixes \<sigma> msg dests sip
+  assumes "rerr_invalid (rt (\<sigma> sip)) (Rerr dests sip)"
+  shows "msg_fresh \<sigma> (Rerr dests sip)"
+        (is "msg_fresh \<sigma> ?msg")
+  proof -
+    let ?rt = "rt (\<sigma> sip)"
+    from assms have *: "(\<forall>rip\<in>dom(dests). (rip\<in>iD(rt (\<sigma> sip))
+                                     \<and> the (dests rip) = sqn (rt (\<sigma> sip)) rip))"
+      by clarsimp
+    have "(\<forall>rip\<in>dom(dests). (rip\<in>kD(rt (\<sigma> sip))
+                                     \<and> the (dests rip) - 1 \<le> nsqn (rt (\<sigma> sip)) rip))"
+    proof
+      fix rip
+      assume "rip \<in> dom dests"
+      with * have "rip\<in>iD(rt (\<sigma> sip))" and "the (dests rip) = sqn (rt (\<sigma> sip)) rip"
+        by auto
+
+      from this(2) have "the (dests rip) - 1 = sqn (rt (\<sigma> sip)) rip - 1" by simp
+      also have "... \<le> nsqn (rt (\<sigma> sip)) rip" by (rule sqn_nsqn)
+      finally have "the (dests rip) - 1 \<le> nsqn (rt (\<sigma> sip)) rip" .
+
+      with \<open>rip\<in>iD(rt (\<sigma> sip))\<close>
+        show "rip\<in>kD(rt (\<sigma> sip)) \<and> the (dests rip) - 1 \<le> nsqn (rt (\<sigma> sip)) rip"
+          by clarsimp
+    qed
+    thus "msg_fresh \<sigma> ?msg"
+      by simp
+  qed
+
+lemma quality_increases_msg_fresh [elim]:
+  assumes qinc: "\<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)"
+      and "msg_fresh \<sigma> m"
+    shows "msg_fresh \<sigma>' m"
+  using assms(2)
+  proof (cases m)
+    fix hops rreqid dip dsn dsk oip osn sip
+    assume [simp]: "m = Rreq hops  dip dsn dsk oip osn sip"
+       and "msg_fresh \<sigma> m"
+    then have "osn \<ge> 1" and "sip = oip \<or> (oip\<in>kD(rt (\<sigma> sip)) \<and> osn \<le> nsqn (rt (\<sigma> sip)) oip
+                                           \<and> (nsqn (rt (\<sigma> sip)) oip = osn
+                                                 \<longrightarrow> (the (dhops (rt (\<sigma> sip)) oip) \<le> hops
+                                                      \<or> the (flag (rt (\<sigma> sip)) oip) = inv)))"
+      by auto
+    from this(2) show ?thesis
+    proof
+      assume "sip = oip" with \<open>osn \<ge> 1\<close> show ?thesis by simp
+    next
+      assume "oip\<in>kD(rt (\<sigma> sip)) \<and> osn \<le> nsqn (rt (\<sigma> sip)) oip
+                                  \<and> (nsqn (rt (\<sigma> sip)) oip = osn
+                                      \<longrightarrow> (the (dhops (rt (\<sigma> sip)) oip) \<le> hops
+                                           \<or> the (flag (rt (\<sigma> sip)) oip) = inv))"
+      moreover from qinc have "quality_increases (\<sigma> sip) (\<sigma>' sip)" ..
+      ultimately have "oip\<in>kD(rt (\<sigma>' sip)) \<and> osn \<le> nsqn (rt (\<sigma>' sip)) oip
+                                           \<and> (nsqn (rt (\<sigma>' sip)) oip = osn
+                                              \<longrightarrow> (the (dhops (rt (\<sigma>' sip)) oip) \<le> hops
+                                                    \<or> the (flag (rt (\<sigma>' sip)) oip) = inv))"
+       using \<open>osn \<ge> 1\<close> by (rule quality_increases_rreq_rrep_props [rotated 2])
+      with \<open>osn \<ge> 1\<close> show "msg_fresh \<sigma>' m"
+        by (clarsimp)
+    qed
+  next
+    fix hops dip dsn oip sip
+    assume [simp]: "m = Rrep hops dip dsn oip sip"
+       and "msg_fresh \<sigma> m"
+    then have "dsn \<ge> 1" and "sip = dip \<or> (dip\<in>kD(rt (\<sigma> sip)) \<and> dsn \<le> nsqn (rt (\<sigma> sip)) dip
+                                           \<and> (nsqn (rt (\<sigma> sip)) dip = dsn
+                                                 \<longrightarrow> (the (dhops (rt (\<sigma> sip)) dip) \<le> hops
+                                                      \<or> the (flag (rt (\<sigma> sip)) dip) = inv)))"
+      by auto
+    from this(2) show "?thesis"
+    proof
+      assume "sip = dip" with \<open>dsn \<ge> 1\<close> show ?thesis by simp
+    next
+      assume "dip\<in>kD(rt (\<sigma> sip)) \<and> dsn \<le> nsqn (rt (\<sigma> sip)) dip
+                                  \<and> (nsqn (rt (\<sigma> sip)) dip = dsn
+                                      \<longrightarrow> (the (dhops (rt (\<sigma> sip)) dip) \<le> hops
+                                           \<or> the (flag (rt (\<sigma> sip)) dip) = inv))"
+      moreover from qinc have "quality_increases (\<sigma> sip) (\<sigma>' sip)" ..
+      ultimately have "dip\<in>kD(rt (\<sigma>' sip)) \<and> dsn \<le> nsqn (rt (\<sigma>' sip)) dip
+                                           \<and> (nsqn (rt (\<sigma>' sip)) dip = dsn
+                                              \<longrightarrow> (the (dhops (rt (\<sigma>' sip)) dip) \<le> hops
+                                                    \<or> the (flag (rt (\<sigma>' sip)) dip) = inv))"
+        using \<open>dsn \<ge> 1\<close> by (rule quality_increases_rreq_rrep_props [rotated 2])
+      with \<open>dsn \<ge> 1\<close> show "msg_fresh \<sigma>' m"
+        by clarsimp
+    qed
+  next
+    fix dests sip
+    assume [simp]: "m = Rerr dests sip"
+       and "msg_fresh \<sigma> m"
+    then have *: "\<forall>rip\<in>dom(dests). rip\<in>kD(rt (\<sigma> sip))
+                              \<and> the (dests rip) - 1 \<le> nsqn (rt (\<sigma> sip)) rip"
+      by simp
+    have "\<forall>rip\<in>dom(dests). rip\<in>kD(rt (\<sigma>' sip))
+                         \<and> the (dests rip) - 1 \<le> nsqn (rt (\<sigma>' sip)) rip"
+      proof
+        fix rip
+        assume "rip\<in>dom(dests)"
+        with * have "rip\<in>kD(rt (\<sigma> sip))" and "the (dests rip) - 1 \<le> nsqn (rt (\<sigma> sip)) rip"
+          by - (drule(1) bspec, clarsimp)+
+        moreover from qinc have "quality_increases (\<sigma> sip) (\<sigma>' sip)" by simp
+        ultimately show "rip\<in>kD(rt (\<sigma>' sip)) \<and> the (dests rip) - 1 \<le> nsqn (rt (\<sigma>' sip)) rip" ..
+      qed
+    thus ?thesis by simp
+  qed simp_all
+
+end

formal/afp/AODV/variants/a_norreqid/A_Seq_Invariants.thy

@@ -0,0 +1,643 @@
+(*  Title:       variants/a_norreqid/Seq_Invariants.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "Invariant proofs on individual processes"
+
+theory A_Seq_Invariants
+imports AWN.Invariants A_Aodv A_Aodv_Data A_Aodv_Predicates A_Fresher
+
+begin
+
+text \<open>
+  The proposition numbers are taken from the December 2013 version of
+  the Fehnker et al technical report.
+\<close>
+
+text \<open>Proposition 7.2\<close>
+
+lemma sequence_number_increases:
+  "paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). sn \<xi> \<le> sn \<xi>')"
+  by inv_cterms
+
+lemma sequence_number_one_or_bigger:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, _). 1 \<le> sn \<xi>)"
+  by (rule onll_step_to_invariantI [OF sequence_number_increases])
+     (auto simp: \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def)
+
+text \<open>We can get rid of the onl/onll if desired...\<close>
+
+lemma sequence_number_increases':
+  "paodv i \<TTurnstile>\<^sub>A (\<lambda>((\<xi>, _), _, (\<xi>', _)). sn \<xi> \<le> sn \<xi>')"
+  by (rule step_invariant_weakenE [OF sequence_number_increases]) (auto dest!: onllD)
+
+lemma sequence_number_one_or_bigger':
+  "paodv i \<TTurnstile> (\<lambda>(\<xi>, _). 1 \<le> sn \<xi>)"
+  by (rule invariant_weakenE [OF sequence_number_one_or_bigger]) auto
+
+lemma sip_in_kD:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). l \<in> ({PAodv-:7} \<union> {PAodv-:5} \<union> {PRrep-:0..PRrep-:1}
+                                     \<union> {PRreq-:0..PRreq-:3}) \<longrightarrow> sip \<xi> \<in> kD (rt \<xi>))"
+  by inv_cterms
+
+lemma rrep_1_update_changes:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). (l = PRrep-:1 \<longrightarrow>
+                        rt \<xi> \<noteq> update (rt \<xi>) (dip \<xi>) (dsn \<xi>, kno, val, hops \<xi> + 1, sip \<xi>, {})))"
+  by inv_cterms
+
+lemma addpreRT_partly_welldefined:
+  "paodv i \<TTurnstile>
+     onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). (l \<in> {PRreq-:16..PRreq-:18} \<union> {PRrep-:2..PRrep-:6} \<longrightarrow> dip \<xi> \<in> kD (rt \<xi>))
+                      \<and> (l \<in> {PRreq-:3..PRreq-:17} \<longrightarrow> oip \<xi> \<in> kD (rt \<xi>)))"
+  by inv_cterms
+
+text \<open>Proposition 7.38\<close>
+
+lemma includes_nhip:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). \<forall>dip\<in>kD(rt \<xi>). the (nhop (rt \<xi>) dip)\<in>kD(rt \<xi>))"
+  proof -
+    { fix ip and \<xi> \<xi>' :: state
+      assume "\<forall>dip\<in>kD (rt \<xi>). the (nhop (rt \<xi>) dip) \<in> kD (rt \<xi>)"
+         and "\<xi>' = \<xi>\<lparr>rt := update (rt \<xi>) ip (0, unk, val, Suc 0, ip, {})\<rparr>"
+      hence "\<forall>dip\<in>kD (rt \<xi>).
+               the (nhop (update (rt \<xi>) ip (0, unk, val, Suc 0, ip, {})) dip) = ip
+             \<or> the (nhop (update (rt \<xi>) ip (0, unk, val, Suc 0, ip, {})) dip) \<in> kD (rt \<xi>)"
+        by clarsimp (metis nhop_update_unk_val update_another)
+    } note one_hop = this
+    {  fix ip sip sn hops and \<xi> \<xi>' :: state
+       assume "\<forall>dip\<in>kD (rt \<xi>). the (nhop (rt \<xi>) dip) \<in> kD (rt \<xi>)"
+          and "\<xi>' = \<xi>\<lparr>rt := update (rt \<xi>) ip (sn, kno, val, Suc hops, sip, {})\<rparr>"
+          and "sip \<in> kD (rt \<xi>)"
+       hence "(the (nhop (update (rt \<xi>) ip (sn, kno, val, Suc hops, sip, {})) ip) = ip
+                 \<or> the (nhop (update (rt \<xi>) ip (sn, kno, val, Suc hops, sip, {})) ip) \<in> kD (rt \<xi>))
+               \<and> (\<forall>dip\<in>kD (rt \<xi>).
+                    the (nhop (update (rt \<xi>) ip (sn, kno, val, Suc hops, sip, {})) dip) = ip
+                    \<or> the (nhop (update (rt \<xi>) ip (sn, kno, val, Suc hops, sip, {})) dip) \<in> kD (rt \<xi>))"
+         by (metis kD_update_unchanged nhop_update_changed update_another)
+    } note nhip_is_sip = this
+    show ?thesis
+      by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf sip_in_kD]
+                              onl_invariant_sterms [OF aodv_wf addpreRT_partly_welldefined]
+                       solve: one_hop nhip_is_sip)
+  qed
+
+text \<open>Proposition 7.22: needed in Proposition 7.4\<close>
+
+lemma addpreRT_welldefined:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). (l \<in> {PRreq-:16..PRreq-:18} \<longrightarrow> dip \<xi> \<in> kD (rt \<xi>)) \<and>
+                               (l = PRreq-:17 \<longrightarrow> oip \<xi> \<in> kD (rt \<xi>)) \<and>                  
+                               (l = PRrep-:5  \<longrightarrow> dip \<xi> \<in> kD (rt \<xi>)) \<and>
+                               (l = PRrep-:6  \<longrightarrow> (the (nhop (rt \<xi>) (dip \<xi>))) \<in> kD (rt \<xi>)))"
+  (is "_ \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V ?P")
+  unfolding invariant_def
+  proof
+    fix s
+    assume "s \<in> reachable (paodv i) TT"
+    then obtain \<xi> p where "s = (\<xi>, p)"
+                      and "(\<xi>, p) \<in> reachable (paodv i) TT"
+      by (metis prod.exhaust)
+    have "onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V ?P (\<xi>, p)"
+    proof (rule onlI)
+      fix l
+      assume "l \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p"
+      with \<open>(\<xi>, p) \<in> reachable (paodv i) TT\<close>
+        have I1: "l \<in> {PRreq-:16..PRreq-:18} \<longrightarrow> dip \<xi> \<in> kD(rt \<xi>)"
+         and I2: "l = PRreq-:17 \<longrightarrow> oip \<xi> \<in> kD(rt \<xi>)"
+         and I3: "l \<in> {PRrep-:2..PRrep-:6}  \<longrightarrow> dip \<xi> \<in> kD(rt \<xi>)"
+         by (auto dest!: invariantD [OF addpreRT_partly_welldefined])
+      moreover from \<open>(\<xi>, p) \<in> reachable (paodv i) TT\<close> \<open>l \<in> labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p\<close> and I3
+        have "l = PRrep-:6  \<longrightarrow> (the (nhop (rt \<xi>) (dip \<xi>))) \<in> kD(rt \<xi>)"
+          by (auto dest!: invariantD [OF includes_nhip])
+      ultimately show "?P (\<xi>, l)"
+        by simp
+    qed
+    with \<open>s = (\<xi>, p)\<close> show "onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V ?P s"
+      by simp
+  qed
+
+text \<open>Proposition 7.4\<close>
+
+lemma known_destinations_increase:
+  "paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). kD (rt \<xi>) \<subseteq> kD (rt \<xi>'))"
+  by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf addpreRT_welldefined]
+                 simp add: subset_insertI)
+
+text \<open>Proposition 7.5\<close>
+
+lemma rreqs_increase:
+  "paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). rreqs \<xi> \<subseteq> rreqs \<xi>')"
+  by (inv_cterms simp add: subset_insertI)
+
+lemma dests_bigger_than_sqn:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). l \<in> {PAodv-:15..PAodv-:19}
+                                 \<union> {PPkt-:7..PPkt-:11}
+                                 \<union> {PRreq-:9..PRreq-:13}
+                                 \<union> {PRreq-:21..PRreq-:25}
+                                 \<union> {PRrep-:10..PRrep-:14}
+                                 \<union> {PRerr-:1..PRerr-:5}
+                         \<longrightarrow> (\<forall>ip\<in>dom(dests \<xi>). ip\<in>kD(rt \<xi>) \<and> sqn (rt \<xi>) ip \<le> the (dests \<xi> ip)))"
+  proof -
+    have sqninv:
+      "\<And>dests rt rsn ip.
+       \<lbrakk> \<forall>ip\<in>dom(dests). ip\<in>kD(rt) \<and> sqn rt ip \<le> the (dests ip); dests ip = Some rsn \<rbrakk>
+        \<Longrightarrow> sqn (invalidate rt dests) ip \<le> rsn"
+        by (rule sqn_invalidate_in_dests [THEN eq_imp_le], assumption) auto
+    have indests:
+      "\<And>dests rt rsn ip.
+       \<lbrakk> \<forall>ip\<in>dom(dests). ip\<in>kD(rt) \<and> sqn rt ip \<le> the (dests ip); dests ip = Some rsn \<rbrakk>
+        \<Longrightarrow> ip\<in>kD(rt) \<and> sqn rt ip \<le> rsn"
+        by (metis domI option.sel)
+    show ?thesis
+      by inv_cterms
+         (clarsimp split: if_split_asm option.split_asm
+                   elim!: sqninv indests)+
+  qed
+
+text \<open>Proposition 7.6\<close>
+
+lemma sqns_increase:
+   "paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)). \<forall>ip. sqn (rt \<xi>) ip \<le> sqn (rt \<xi>') ip)"
+  proof -
+    { fix \<xi> :: state
+      assume *: "\<forall>ip\<in>dom(dests \<xi>). ip \<in> kD (rt \<xi>) \<and> sqn (rt \<xi>) ip \<le> the (dests \<xi> ip)"
+      have "\<forall>ip. sqn (rt \<xi>) ip \<le> sqn (invalidate (rt \<xi>) (dests \<xi>)) ip"
+      proof
+        fix ip
+        from * have "ip\<notin>dom(dests \<xi>) \<or> sqn (rt \<xi>) ip \<le> the (dests \<xi> ip)" by simp
+        thus "sqn (rt \<xi>) ip \<le> sqn (invalidate (rt \<xi>) (dests \<xi>)) ip"
+          by (metis domI invalidate_sqn option.sel)
+      qed
+    } note solve_invalidate = this
+    show ?thesis
+      by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf addpreRT_welldefined]
+                              onl_invariant_sterms [OF aodv_wf dests_bigger_than_sqn]
+                    simp add: solve_invalidate)
+  qed
+
+text \<open>Proposition 7.7\<close>
+
+lemma ip_constant:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, _). ip \<xi> = i)"
+  by (inv_cterms simp add: \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def)
+
+text \<open>Proposition 7.8\<close>
+
+lemma sender_ip_valid':
+  "paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), a, _). anycast (\<lambda>m. not_Pkt m \<longrightarrow> msg_sender m = ip \<xi>) a)"
+  by inv_cterms
+
+lemma sender_ip_valid:
+  "paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), a, _). anycast (\<lambda>m. not_Pkt m \<longrightarrow> msg_sender m = i) a)"
+  by (rule step_invariant_weaken_with_invariantE [OF ip_constant sender_ip_valid'])
+     (auto dest!: onlD onllD)
+
+lemma received_msg_inv:
+  "paodv i \<TTurnstile> (recvmsg P \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). l \<in> {PAodv-:1} \<longrightarrow> P (msg \<xi>))"
+  by inv_cterms
+
+text \<open>Proposition 7.9\<close>
+
+lemma sip_not_ip':
+  "paodv i \<TTurnstile> (recvmsg (\<lambda>m. not_Pkt m \<longrightarrow> msg_sender m \<noteq> i) \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, _). sip \<xi> \<noteq> ip \<xi>)"
+  by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf received_msg_inv]
+                          onl_invariant_sterms [OF aodv_wf ip_constant [THEN invariant_restrict_inD]]
+                simp add: clear_locals_sip_not_ip') clarsimp+
+
+lemma sip_not_ip:
+  "paodv i \<TTurnstile> (recvmsg (\<lambda>m. not_Pkt m \<longrightarrow> msg_sender m \<noteq> i) \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, _). sip \<xi> \<noteq> i)"
+  by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf received_msg_inv]
+                          onl_invariant_sterms [OF aodv_wf ip_constant [THEN invariant_restrict_inD]]
+                simp add: clear_locals_sip_not_ip') clarsimp+
+
+text \<open>Neither \<open>sip_not_ip'\<close> nor \<open>sip_not_ip\<close> is needed to show loop freedom.\<close>
+
+text \<open>Proposition 7.10\<close>
+
+lemma hop_count_positive:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, _). \<forall>ip\<in>kD (rt \<xi>). the (dhops (rt \<xi>) ip) \<ge> 1)"
+  by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf addpreRT_welldefined]) auto
+
+lemma rreq_dip_in_vD_dip_eq_ip:
+  "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). (l \<in> {PRreq-:16..PRreq-:18} \<longrightarrow> dip \<xi> \<in> vD(rt \<xi>))
+                            \<and> (l \<in> {PRreq-:5, PRreq-:6} \<longrightarrow> dip \<xi> = ip \<xi>)
+                            \<and> (l \<in> {PRreq-:15..PRreq-:18} \<longrightarrow> dip \<xi> \<noteq> ip \<xi>))"
+  proof (inv_cterms, elim conjE)
+    fix l \<xi> pp p'
+    assume "(\<xi>, pp) \<in> reachable (paodv i) TT"
+       and "{PRreq-:17}\<lbrakk>\<lambda>\<xi>. \<xi>\<lparr>rt := the (addpreRT (rt \<xi>) (oip \<xi>) {the (nhop (rt \<xi>) (dip \<xi>))})\<rparr>\<rbrakk> p'
+              \<in> sterms \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pp"
+       and "l = PRreq-:17"
+       and "dip \<xi> \<in> vD (rt \<xi>)"
+    from this(1-3) have "oip \<xi> \<in> kD (rt \<xi>)"
+      by (auto dest: onl_invariant_sterms [OF aodv_wf addpreRT_welldefined, where l="PRreq-:17"])
+    with \<open>dip \<xi> \<in> vD (rt \<xi>)\<close>
+      show "dip \<xi> \<in> vD (the (addpreRT (rt \<xi>) (oip \<xi>) {the (nhop (rt \<xi>) (dip \<xi>))}))" by simp
+  qed
+
+text \<open>Proposition 7.11\<close>
+
+lemma anycast_msg_zhops:
+  "\<And>rreqid dip dsn dsk oip osn sip.
+      paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(_, a, _). anycast msg_zhops a)"
+  proof (inv_cterms inv add:
+           onl_invariant_sterms [OF aodv_wf rreq_dip_in_vD_dip_eq_ip [THEN invariant_restrict_inD]]
+           onl_invariant_sterms [OF aodv_wf hop_count_positive [THEN invariant_restrict_inD]],
+         elim conjE)
+    fix l \<xi> a pp p' pp'
+    assume "(\<xi>, pp) \<in> reachable (paodv i) TT"
+       and "{PRreq-:18}unicast(\<lambda>\<xi>. the (nhop (rt \<xi>) (oip \<xi>)),
+               \<lambda>\<xi>. Rrep (the (dhops (rt \<xi>) (dip \<xi>))) (dip \<xi>) (sqn (rt \<xi>) (dip \<xi>)) (oip \<xi>) (ip \<xi>)).
+                     p' \<triangleright> pp' \<in> sterms \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pp"
+       and "l = PRreq-:18"
+       and "a = unicast (the (nhop (rt \<xi>) (oip \<xi>)))
+                 (Rrep (the (dhops (rt \<xi>) (dip \<xi>))) (dip \<xi>) (sqn (rt \<xi>) (dip \<xi>)) (oip \<xi>) (ip \<xi>))"
+       and *: "\<forall>ip\<in>kD (rt \<xi>). Suc 0 \<le> the (dhops (rt \<xi>) ip)"
+       and "dip \<xi> \<in> vD (rt \<xi>)"
+    from \<open>dip \<xi> \<in> vD (rt \<xi>)\<close> have "dip \<xi> \<in> kD (rt \<xi>)"
+      by (rule vD_iD_gives_kD(1))
+    with * have "Suc 0 \<le> the (dhops (rt \<xi>) (dip \<xi>))" ..
+    thus "0 < the (dhops (rt \<xi>) (dip \<xi>))" by simp
+  qed
+
+lemma hop_count_zero_oip_dip_sip:
+  "paodv i \<TTurnstile> (recvmsg msg_zhops \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                 (l\<in>{PAodv-:4..PAodv-:5} \<union> {PRreq-:n|n. True} \<longrightarrow>
+                          (hops \<xi> = 0 \<longrightarrow> oip \<xi> = sip \<xi>))
+                 \<and>
+                 ((l\<in>{PAodv-:6..PAodv-:7} \<union> {PRrep-:n|n. True} \<longrightarrow>
+                          (hops \<xi> = 0 \<longrightarrow> dip \<xi> = sip \<xi>))))"
+  by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf received_msg_inv]) auto
+
+lemma osn_rreq:
+  "paodv i \<TTurnstile> (recvmsg rreq_rrep_sn \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                                    l \<in> {PAodv-:4, PAodv-:5} \<union> {PRreq-:n|n. True} \<longrightarrow> 1 \<le> osn \<xi>)"
+  by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf received_msg_inv]) clarsimp
+
+lemma osn_rreq':
+  "paodv i \<TTurnstile> (recvmsg (\<lambda>m. rreq_rrep_sn m \<and> msg_zhops m) \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                                    l \<in> {PAodv-:4, PAodv-:5} \<union> {PRreq-:n|n. True} \<longrightarrow> 1 \<le> osn \<xi>)"
+  proof (rule invariant_weakenE [OF osn_rreq])
+    fix a
+    assume "recvmsg (\<lambda>m. rreq_rrep_sn m \<and> msg_zhops m) a"
+    thus "recvmsg rreq_rrep_sn a"
+      by (cases a) simp_all
+  qed
+
+lemma dsn_rrep:
+  "paodv i \<TTurnstile> (recvmsg rreq_rrep_sn \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                                    l \<in> {PAodv-:6, PAodv-:7} \<union> {PRrep-:n|n. True} \<longrightarrow> 1 \<le> dsn \<xi>)"
+  by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf received_msg_inv]) clarsimp
+
+lemma dsn_rrep':
+  "paodv i \<TTurnstile> (recvmsg (\<lambda>m. rreq_rrep_sn m \<and> msg_zhops m) \<rightarrow>)  onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                                    l \<in> {PAodv-:6, PAodv-:7} \<union> {PRrep-:n|n. True} \<longrightarrow> 1 \<le> dsn \<xi>)"
+  proof (rule invariant_weakenE [OF dsn_rrep])
+    fix a
+    assume "recvmsg (\<lambda>m. rreq_rrep_sn m \<and> msg_zhops m) a"
+    thus "recvmsg rreq_rrep_sn a"
+      by (cases a) simp_all
+  qed
+
+lemma hop_count_zero_oip_dip_sip':
+  "paodv i \<TTurnstile> (recvmsg (\<lambda>m. rreq_rrep_sn m \<and> msg_zhops m) \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                 (l\<in>{PAodv-:4..PAodv-:5} \<union> {PRreq-:n|n. True} \<longrightarrow>
+                          (hops \<xi> = 0 \<longrightarrow> oip \<xi> = sip \<xi>))
+                 \<and>
+                 ((l\<in>{PAodv-:6..PAodv-:7} \<union> {PRrep-:n|n. True} \<longrightarrow>
+                          (hops \<xi> = 0 \<longrightarrow> dip \<xi> = sip \<xi>))))"
+  proof (rule invariant_weakenE [OF hop_count_zero_oip_dip_sip])
+    fix a
+    assume "recvmsg (\<lambda>m. rreq_rrep_sn m \<and> msg_zhops m) a"
+    thus "recvmsg msg_zhops a"
+      by (cases a) simp_all
+  qed
+
+text \<open>Proposition 7.12\<close>
+
+lemma zero_seq_unk_hops_one':
+  "paodv i \<TTurnstile> (recvmsg (\<lambda>m. rreq_rrep_sn m \<and> msg_zhops m) \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, _).
+                 \<forall>dip\<in>kD(rt \<xi>). (sqn (rt \<xi>) dip = 0 \<longrightarrow> sqnf (rt \<xi>) dip = unk)
+                              \<and> (sqnf (rt \<xi>) dip = unk \<longrightarrow> the (dhops (rt \<xi>) dip) = 1)
+                              \<and> (the (dhops (rt \<xi>) dip) = 1 \<longrightarrow> the (nhop (rt \<xi>) dip) = dip))"
+  proof -
+  { fix dip and \<xi> :: state and P
+    assume "sqn (invalidate (rt \<xi>) (dests \<xi>)) dip = 0"
+       and all: "\<forall>ip. sqn (rt \<xi>) ip \<le> sqn (invalidate (rt \<xi>) (dests \<xi>)) ip"
+       and *: "sqn (rt \<xi>) dip = 0 \<Longrightarrow> P \<xi> dip"
+    have "P \<xi> dip"
+    proof -
+      from all have "sqn (rt \<xi>) dip \<le> sqn (invalidate (rt \<xi>) (dests \<xi>)) dip" ..
+      with \<open>sqn (invalidate (rt \<xi>) (dests \<xi>)) dip = 0\<close> have "sqn (rt \<xi>) dip = 0" by simp
+      thus "P \<xi> dip" by (rule *)
+    qed
+  } note sqn_invalidate_zero [elim!] = this
+
+  { fix dsn hops :: nat and sip oip rt and ip dip :: ip
+    assume "\<forall>dip\<in>kD(rt).
+                (sqn rt dip = 0 \<longrightarrow> \<pi>\<^sub>3(the (rt dip)) = unk) \<and>
+                (\<pi>\<^sub>3(the (rt dip)) = unk \<longrightarrow> the (dhops rt dip) = Suc 0) \<and>
+                (the (dhops rt dip) = Suc 0 \<longrightarrow> the (nhop rt dip) = dip)"
+       and "hops = 0 \<longrightarrow> sip = dip"
+       and "Suc 0 \<le> dsn"
+       and "ip \<noteq> dip \<longrightarrow> ip\<in>kD(rt)"
+    hence "the (dhops (update rt dip (dsn, kno, val, Suc hops, sip, {})) ip) = Suc 0 \<longrightarrow>
+           the (nhop (update rt dip (dsn, kno, val, Suc hops, sip, {})) ip) = ip"
+      by - (rule update_cases, auto simp add: sqn_def dest!: bspec)
+  } note prreq_ok1 [simp] = this
+
+  { fix ip dsn hops sip oip rt dip
+    assume "\<forall>dip\<in>kD(rt).
+                (sqn rt dip = 0 \<longrightarrow> \<pi>\<^sub>3(the (rt dip)) = unk) \<and>
+                (\<pi>\<^sub>3(the (rt dip)) = unk \<longrightarrow> the (dhops rt dip) = Suc 0) \<and>
+                (the (dhops rt dip) = Suc 0 \<longrightarrow> the (nhop rt dip) = dip)"
+       and "Suc 0 \<le> dsn"
+       and "ip \<noteq> dip \<longrightarrow> ip\<in>kD(rt)"
+    hence "\<pi>\<^sub>3(the (update rt dip (dsn, kno, val, Suc hops, sip, {}) ip)) = unk \<longrightarrow>
+           the (dhops (update rt dip (dsn, kno, val, Suc hops, sip, {})) ip) = Suc 0"
+      by - (rule update_cases, auto simp add: sqn_def sqnf_def dest!: bspec)
+  } note prreq_ok2 [simp] = this
+
+  { fix ip dsn hops sip oip rt dip
+    assume "\<forall>dip\<in>kD(rt).
+                (sqn rt dip = 0 \<longrightarrow> \<pi>\<^sub>3(the (rt dip)) = unk) \<and>
+                (\<pi>\<^sub>3(the (rt dip)) = unk \<longrightarrow> the (dhops rt dip) = Suc 0) \<and>
+                (the (dhops rt dip) = Suc 0 \<longrightarrow> the (nhop rt dip) = dip)"
+       and "Suc 0 \<le> dsn"
+       and "ip \<noteq> dip \<longrightarrow> ip\<in>kD(rt)"
+    hence "sqn (update rt dip (dsn, kno, val, Suc hops, sip, {})) ip = 0 \<longrightarrow>
+           \<pi>\<^sub>3 (the (update rt dip (dsn, kno, val, Suc hops, sip, {}) ip)) = unk"
+      by - (rule update_cases, auto simp add: sqn_def dest!: bspec)
+  } note prreq_ok3 [simp] = this
+
+  { fix rt sip
+    assume "\<forall>dip\<in>kD rt.
+              (sqn rt dip = 0 \<longrightarrow> \<pi>\<^sub>3(the (rt dip)) = unk) \<and>
+              (\<pi>\<^sub>3(the (rt dip)) = unk \<longrightarrow> the (dhops rt dip) = Suc 0) \<and>
+              (the (dhops rt dip) = Suc 0 \<longrightarrow> the (nhop rt dip) = dip)"
+    hence "\<forall>dip\<in>kD rt.
+          (sqn (update rt sip (0, unk, val, Suc 0, sip, {})) dip = 0 \<longrightarrow>
+           \<pi>\<^sub>3(the (update rt sip (0, unk, val, Suc 0, sip, {}) dip)) = unk)
+        \<and> (\<pi>\<^sub>3(the (update rt sip (0, unk, val, Suc 0, sip, {}) dip)) = unk \<longrightarrow>
+           the (dhops (update rt sip (0, unk, val, Suc 0, sip, {})) dip) = Suc 0)
+        \<and> (the (dhops (update rt sip (0, unk, val, Suc 0, sip, {})) dip) = Suc 0 \<longrightarrow>
+           the (nhop (update rt sip (0, unk, val, Suc 0, sip, {})) dip) = dip)"
+    by - (rule update_cases, simp_all add: sqnf_def sqn_def)
+  } note prreq_ok4 [simp] = this
+
+  have prreq_ok5 [simp]: "\<And>sip rt.
+    \<pi>\<^sub>3(the (update rt sip (0, unk, val, Suc 0, sip, {}) sip)) = unk \<longrightarrow>
+    the (dhops (update rt sip (0, unk, val, Suc 0, sip, {})) sip) = Suc 0"
+    by (rule update_cases) simp_all
+
+  have prreq_ok6 [simp]: "\<And>sip rt.
+    sqn (update rt sip (0, unk, val, Suc 0, sip, {})) sip = 0 \<longrightarrow>
+    \<pi>\<^sub>3 (the (update rt sip (0, unk, val, Suc 0, sip, {}) sip)) = unk"
+    by (rule update_cases) simp_all
+
+  show ?thesis
+    by (inv_cterms inv add: onl_invariant_sterms_TT [OF aodv_wf addpreRT_welldefined]
+                            onl_invariant_sterms [OF aodv_wf hop_count_zero_oip_dip_sip']
+                            seq_step_invariant_sterms_TT [OF sqns_increase aodv_wf aodv_trans]
+                            onl_invariant_sterms [OF aodv_wf osn_rreq']
+                            onl_invariant_sterms [OF aodv_wf dsn_rrep']) clarsimp+
+  qed
+
+lemma zero_seq_unk_hops_one:
+  "paodv i \<TTurnstile> (recvmsg (\<lambda>m. rreq_rrep_sn m \<and> msg_zhops m) \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, _).
+                 \<forall>dip\<in>kD(rt \<xi>). (sqn (rt \<xi>) dip = 0 \<longrightarrow> (sqnf (rt \<xi>) dip = unk
+                                                         \<and> the (dhops (rt \<xi>) dip) = 1
+                                                         \<and> the (nhop (rt \<xi>) dip) = dip)))"
+  by (rule invariant_weakenE [OF zero_seq_unk_hops_one']) auto
+
+lemma kD_unk_or_atleast_one:
+  "paodv i \<TTurnstile> (recvmsg rreq_rrep_sn \<rightarrow>) onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                               \<forall>dip\<in>kD(rt \<xi>). \<pi>\<^sub>3(the (rt \<xi> dip)) = unk \<or> 1 \<le> \<pi>\<^sub>2(the (rt \<xi> dip)))"
+  proof -
+    { fix sip rt dsn1 dsn2 dsk1 dsk2 flag1 flag2 hops1 hops2 nhip1 nhip2 pre1 pre2
+      assume "dsk1 = unk \<or> Suc 0 \<le> dsn2"
+      hence "\<pi>\<^sub>3(the (update rt sip (dsn1, dsk1, flag1, hops1, nhip1, pre1) sip)) = unk
+            \<or> Suc 0 \<le> sqn (update rt sip (dsn2, dsk2, flag2, hops2, nhip2, pre2)) sip"
+        unfolding update_def by (cases "dsk1 =unk") (clarsimp split: option.split)+
+    } note fromsip [simp] = this
+
+    { fix dip sip rt dsn1 dsn2 dsk1 dsk2 flag1 flag2 hops1 hops2 nhip1 nhip2 pre1 pre2
+      assume allkd: "\<forall>dip\<in>kD(rt). \<pi>\<^sub>3(the (rt dip)) = unk \<or> Suc 0 \<le> sqn rt dip"
+         and    **: "dsk1 = unk \<or> Suc 0 \<le> dsn2"
+      have "\<forall>dip\<in>kD(rt). \<pi>\<^sub>3(the (update rt sip (dsn1, dsk1, flag1, hops1, nhip1, pre1) dip)) = unk
+            \<or> Suc 0 \<le> sqn (update rt sip (dsn2, dsk2, flag2, hops2, nhip2, pre2)) dip"
+        (is "\<forall>dip\<in>kD(rt). ?prop dip")
+      proof
+        fix dip
+        assume "dip\<in>kD(rt)"
+        thus "?prop dip"
+        proof (cases "dip = sip")
+          assume "dip = sip"
+          with ** show ?thesis
+            by simp
+        next
+          assume "dip \<noteq> sip"
+          with \<open>dip\<in>kD(rt)\<close> allkd show ?thesis
+            by simp
+        qed
+      qed
+    } note solve_update [simp] = this
+
+    { fix dip rt dests
+      assume  *: "\<forall>ip\<in>dom(dests). ip\<in>kD(rt) \<and> sqn rt ip \<le> the (dests ip)"
+         and **: "\<forall>ip\<in>kD(rt). \<pi>\<^sub>3(the (rt ip)) = unk \<or> Suc 0 \<le> sqn rt ip"
+      have "\<forall>dip\<in>kD(rt). \<pi>\<^sub>3(the (rt dip)) = unk \<or> Suc 0 \<le> sqn (invalidate rt dests) dip"
+      proof
+        fix dip
+        assume "dip\<in>kD(rt)"
+        with ** have "\<pi>\<^sub>3(the (rt dip)) = unk \<or> Suc 0 \<le> sqn rt dip" ..
+        thus "\<pi>\<^sub>3 (the (rt dip)) = unk \<or> Suc 0 \<le> sqn (invalidate rt dests) dip"
+        proof
+          assume "\<pi>\<^sub>3(the (rt dip)) = unk" thus ?thesis ..
+        next
+          assume "Suc 0 \<le> sqn rt dip"
+          have "Suc 0 \<le> sqn (invalidate rt dests) dip"
+          proof (cases "dip\<in>dom(dests)")
+            assume "dip\<in>dom(dests)"
+            with * have "sqn rt dip \<le> the (dests dip)" by simp
+            with \<open>Suc 0 \<le> sqn rt dip\<close> have "Suc 0 \<le> the (dests dip)" by simp
+            with \<open>dip\<in>dom(dests)\<close> \<open>dip\<in>kD(rt)\<close> [THEN kD_Some] show ?thesis
+              unfolding invalidate_def sqn_def by auto
+          next
+            assume "dip\<notin>dom(dests)"
+            with \<open>Suc 0 \<le> sqn rt dip\<close> \<open>dip\<in>kD(rt)\<close> [THEN kD_Some] show ?thesis
+              unfolding invalidate_def sqn_def by auto
+          qed
+        thus ?thesis by (rule disjI2)
+        qed
+      qed
+    } note solve_invalidate [simp] = this
+
+    show ?thesis
+      by (inv_cterms inv add: onl_invariant_sterms_TT [OF aodv_wf addpreRT_welldefined]
+                              onl_invariant_sterms [OF aodv_wf dests_bigger_than_sqn
+                                                               [THEN invariant_restrict_inD]]
+                              onl_invariant_sterms [OF aodv_wf osn_rreq]
+                              onl_invariant_sterms [OF aodv_wf dsn_rrep]
+                    simp add: proj3_inv proj2_eq_sqn)
+  qed
+
+text \<open>Proposition 7.13\<close>
+
+lemma rreq_rrep_sn_any_step_invariant:
+  "paodv i \<TTurnstile>\<^sub>A (recvmsg rreq_rrep_sn \<rightarrow>) onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(_, a, _). anycast rreq_rrep_sn a)"
+  proof -
+    have sqnf_kno: "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                                      (l \<in> {PRreq-:16..PRreq-:18} \<longrightarrow> sqnf (rt \<xi>) (dip \<xi>) = kno))"
+      by (inv_cterms inv add: onl_invariant_sterms_TT [OF aodv_wf addpreRT_welldefined])
+    show ?thesis
+      by (inv_cterms inv add: onl_invariant_sterms_TT [OF aodv_wf addpreRT_welldefined]
+                              onl_invariant_sterms [OF aodv_wf sequence_number_one_or_bigger
+                                                               [THEN invariant_restrict_inD]]
+                              onl_invariant_sterms [OF aodv_wf kD_unk_or_atleast_one]
+                              onl_invariant_sterms_TT [OF aodv_wf sqnf_kno]
+                              onl_invariant_sterms [OF aodv_wf osn_rreq]
+                              onl_invariant_sterms [OF aodv_wf dsn_rrep])
+         (auto simp: proj2_eq_sqn)
+  qed
+
+text \<open>Proposition 7.14\<close>
+
+lemma rreq_rrep_fresh_any_step_invariant:
+  "paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), a, _). anycast (rreq_rrep_fresh (rt \<xi>)) a)"
+  proof -                                                      
+    have rreq_oip: "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+                       (l \<in> {PRreq-:3, PRreq-:4, PRreq-:15, PRreq-:27}
+                               \<longrightarrow> oip \<xi> \<in> kD(rt \<xi>)
+                                 \<and> (sqn (rt \<xi>) (oip \<xi>) > (osn \<xi>)
+                                     \<or> (sqn (rt \<xi>) (oip \<xi>) = (osn \<xi>)
+                                        \<and> the (dhops (rt \<xi>) (oip \<xi>)) \<le> Suc (hops \<xi>)
+                                        \<and> the (flag (rt \<xi>) (oip \<xi>)) = val))))"
+      proof inv_cterms
+        fix l \<xi> l' pp p'
+        assume "(\<xi>, pp) \<in> reachable (paodv i) TT"
+           and "{PRreq-:2}\<lbrakk>\<lambda>\<xi>. \<xi>\<lparr>rt :=
+                update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})\<rparr>\<rbrakk> p' \<in> sterms \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pp"
+           and "l' = PRreq-:3"
+        show "osn \<xi> < sqn (update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})) (oip \<xi>)
+           \<or> (sqn (update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})) (oip \<xi>) = osn \<xi>
+             \<and> the (dhops (update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})) (oip \<xi>))
+                                                                            \<le> Suc (hops \<xi>)
+             \<and> the (flag (update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})) (oip \<xi>))
+                                                                            = val)"
+          unfolding update_def by (clarsimp split: option.split)
+                                  (metis linorder_neqE_nat not_less)
+      qed
+
+    have rrep_prrep: "paodv i \<TTurnstile> onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l).
+          (l \<in> {PRrep-:2..PRrep-:7} \<longrightarrow> (dip \<xi> \<in> kD(rt \<xi>)
+                                        \<and> sqn (rt \<xi>) (dip \<xi>) = dsn \<xi>
+                                        \<and> the (dhops (rt \<xi>) (dip \<xi>)) = Suc (hops \<xi>)
+                                        \<and> the (flag (rt \<xi>) (dip \<xi>)) = val
+                                        \<and> the (nhop (rt \<xi>) (dip \<xi>)) \<in> kD (rt \<xi>))))"
+      by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf rrep_1_update_changes]
+                              onl_invariant_sterms [OF aodv_wf sip_in_kD])
+
+    show ?thesis
+      by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf rreq_oip]
+                              onl_invariant_sterms [OF aodv_wf rreq_dip_in_vD_dip_eq_ip]
+                              onl_invariant_sterms [OF aodv_wf rrep_prrep])
+  qed
+
+text \<open>Proposition 7.15\<close>
+
+lemma rerr_invalid_any_step_invariant:
+  "paodv i \<TTurnstile>\<^sub>A onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), a, _). anycast (rerr_invalid (rt \<xi>)) a)"
+  proof -
+    have dests_inv: "paodv i \<TTurnstile>
+                      onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). (l \<in> {PAodv-:15, PPkt-:7, PRreq-:9,
+                                            PRreq-:21, PRrep-:10, PRerr-:1}
+                          \<longrightarrow> (\<forall>ip\<in>dom(dests \<xi>). ip\<in>vD(rt \<xi>)))
+                         \<and> (l \<in> {PAodv-:16..PAodv-:19}
+                              \<union> {PPkt-:8..PPkt-:11}
+                              \<union> {PRreq-:10..PRreq-:13}
+                              \<union> {PRreq-:22..PRreq-:25}
+                              \<union> {PRrep-:11..PRrep-:14}
+                              \<union> {PRerr-:2..PRerr-:5} \<longrightarrow> (\<forall>ip\<in>dom(dests \<xi>). ip\<in>iD(rt \<xi>)
+                                                          \<and> the (dests \<xi> ip) = sqn (rt \<xi>) ip))
+                         \<and> (l = PPkt-:14 \<longrightarrow> dip \<xi>\<in>iD(rt \<xi>)))"
+      by inv_cterms (clarsimp split: if_split_asm option.split_asm simp: domIff)+
+    show ?thesis
+      by (inv_cterms inv add: onl_invariant_sterms [OF aodv_wf dests_inv])
+  qed
+
+text \<open>Proposition 7.16\<close>
+
+text \<open>
+  Some well-definedness obligations are irrelevant for the Isabelle development:
+
+  \begin{enumerate}
+  \item In each routing table there is at most one entry for each destination: guaranteed by type.
+
+  \item In each store of queued data packets there is at most one data queue for
+        each destination: guaranteed by structure.
+
+  \item Whenever a set of pairs @{term "(rip, rsn)"} is assigned to the variable
+        @{term "dests"} of type @{typ "ip \<rightharpoonup> sqn"}, or to the first argument of
+        the function @{term "rerr"}, this set is a partial function, i.e., there
+        is at most one entry @{term "(rip, rsn)"} for each destination
+        @{term "rip"}: guaranteed by type.
+  \end{enumerate}
+\<close>
+
+lemma dests_vD_inc_sqn:
+  "paodv i \<TTurnstile>
+        onl \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(\<xi>, l). (l \<in> {PAodv-:15, PPkt-:7, PRreq-:9, PRreq-:21, PRrep-:10}
+             \<longrightarrow> (\<forall>ip\<in>dom(dests \<xi>). ip\<in>vD(rt \<xi>) \<and> the (dests \<xi> ip) = inc (sqn (rt \<xi>) ip)))
+           \<and> (l = PRerr-:1
+             \<longrightarrow> (\<forall>ip\<in>dom(dests \<xi>). ip\<in>vD(rt \<xi>) \<and> the (dests \<xi> ip) > sqn (rt \<xi>) ip)))"
+  by inv_cterms (clarsimp split: if_split_asm option.split_asm)+
+
+text \<open>Proposition 7.27\<close>
+
+lemma route_tables_fresher:
+  "paodv i \<TTurnstile>\<^sub>A (recvmsg rreq_rrep_sn \<rightarrow>) onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<xi>, _), _, (\<xi>', _)).
+                                                                \<forall>dip\<in>kD(rt \<xi>). rt \<xi> \<sqsubseteq>\<^bsub>dip\<^esub> rt \<xi>')"
+  proof (inv_cterms inv add:
+           onl_invariant_sterms [OF aodv_wf dests_vD_inc_sqn [THEN invariant_restrict_inD]]
+           onl_invariant_sterms [OF aodv_wf hop_count_positive [THEN invariant_restrict_inD]]
+           onl_invariant_sterms [OF aodv_wf osn_rreq]
+           onl_invariant_sterms [OF aodv_wf dsn_rrep]
+           onl_invariant_sterms [OF aodv_wf addpreRT_welldefined [THEN invariant_restrict_inD]])
+    fix \<xi> pp p'
+    assume "(\<xi>, pp) \<in> reachable (paodv i) (recvmsg rreq_rrep_sn)"
+       and "{PRreq-:2}\<lbrakk>\<lambda>\<xi>. \<xi>\<lparr>rt := update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})\<rparr>\<rbrakk>
+               p' \<in> sterms \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pp"
+       and "Suc 0 \<le> osn \<xi>"
+       and *: "\<forall>ip\<in>kD (rt \<xi>). Suc 0 \<le> the (dhops (rt \<xi>) ip)"
+    show "\<forall>ip\<in>kD (rt \<xi>). rt \<xi> \<sqsubseteq>\<^bsub>ip\<^esub> update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})"
+    proof
+      fix ip
+      assume "ip\<in>kD (rt \<xi>)"
+      moreover with * have "1 \<le> the (dhops (rt \<xi>) ip)" by simp
+      moreover from \<open>Suc 0 \<le> osn \<xi>\<close>
+        have "update_arg_wf (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})" ..
+      ultimately show "rt \<xi> \<sqsubseteq>\<^bsub>ip\<^esub> update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})"
+        by (rule rt_fresher_update)
+    qed
+  next
+    fix \<xi> pp p'
+    assume "(\<xi>, pp) \<in> reachable (paodv i) (recvmsg rreq_rrep_sn)"
+       and "{PRrep-:1}\<lbrakk>\<lambda>\<xi>. \<xi>\<lparr>rt := update (rt \<xi>) (dip \<xi>) (dsn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})\<rparr>\<rbrakk>
+            p' \<in> sterms \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pp"
+       and "Suc 0 \<le> dsn \<xi>"
+       and *: "\<forall>ip\<in>kD (rt \<xi>). Suc 0 \<le> the (dhops (rt \<xi>) ip)"
+    show "\<forall>ip\<in>kD (rt \<xi>). rt \<xi> \<sqsubseteq>\<^bsub>ip\<^esub> update (rt \<xi>) (dip \<xi>) (dsn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})"
+    proof
+      fix ip
+      assume "ip\<in>kD (rt \<xi>)"
+      moreover with * have "1 \<le> the (dhops (rt \<xi>) ip)" by simp
+      moreover from \<open>Suc 0 \<le> dsn \<xi>\<close>
+        have "update_arg_wf (dsn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})" ..
+      ultimately show "rt \<xi> \<sqsubseteq>\<^bsub>ip\<^esub> update (rt \<xi>) (dip \<xi>) (dsn \<xi>, kno, val, Suc (hops \<xi>), sip \<xi>, {})"
+        by (rule rt_fresher_update)
+    qed
+  qed
+
+end
+

formal/afp/AODV/variants/b_fwdrreps/B_Aodv.thy

@@ -0,0 +1,532 @@
+(*  Title:       variants/b_fwdrreps/Aodv.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+    Author:      Peter Höfner, NICTA
+*)
+
+section "The AODV protocol"
+
+theory B_Aodv
+imports B_Aodv_Data B_Aodv_Message
+        AWN.AWN_SOS_Labels AWN.AWN_Invariants
+begin
+
+subsection "Data state"
+
+record state =
+  ip    :: "ip"
+  sn    :: "sqn"
+  rt    :: "rt" 
+  rreqs :: "(ip \<times> rreqid) set"
+  store :: "store"
+  (* all locals *)
+  msg    :: "msg"
+  data   :: "data"
+  dests  :: "ip \<rightharpoonup> sqn"
+  pre    :: "ip set"
+  rreqid :: "rreqid"
+  dip    :: "ip"
+  oip    :: "ip"
+  hops   :: "nat"
+  dsn    :: "sqn"
+  dsk    :: "k"
+  osn    :: "sqn"
+  sip    :: "ip"
+
+abbreviation aodv_init :: "ip \<Rightarrow> state"
+where "aodv_init i \<equiv> \<lparr>
+         ip = i,
+         sn = 1,
+         rt = Map.empty,
+         rreqs = {},
+         store = Map.empty,
+
+         msg    = (SOME x. True),
+         data   = (SOME x. True),
+         dests  = (SOME x. True),
+         pre    = (SOME x. True),
+         rreqid = (SOME x. True),
+         dip    = (SOME x. True),
+         oip    = (SOME x. True),
+         hops   = (SOME x. True),
+         dsn    = (SOME x. True),
+         dsk    = (SOME x. True),
+         osn    = (SOME x. True),
+         sip    = (SOME x. x \<noteq> i)
+       \<rparr>"
+
+lemma some_neq_not_eq [simp]: "\<not>((SOME x :: nat. x \<noteq> i) = i)"
+  by (subst some_eq_ex) (metis zero_neq_numeral)
+
+definition clear_locals :: "state \<Rightarrow> state"
+where "clear_locals \<xi> = \<xi> \<lparr>
+    msg    := (SOME x. True),
+    data   := (SOME x. True),
+    dests  := (SOME x. True),
+    pre    := (SOME x. True),
+    rreqid := (SOME x. True),
+    dip    := (SOME x. True),
+    oip    := (SOME x. True),
+    hops   := (SOME x. True),
+    dsn    := (SOME x. True),
+    dsk    := (SOME x. True),
+    osn    := (SOME x. True),
+    sip    := (SOME x. x \<noteq> ip \<xi>)
+  \<rparr>"
+
+lemma clear_locals_sip_not_ip [simp]: "\<not>(sip (clear_locals \<xi>) = ip \<xi>)"
+  unfolding clear_locals_def by simp
+
+lemma clear_locals_but_not_globals [simp]:
+  "ip (clear_locals \<xi>) = ip \<xi>"
+  "sn (clear_locals \<xi>) = sn \<xi>"
+  "rt (clear_locals \<xi>) = rt \<xi>"
+  "rreqs (clear_locals \<xi>) = rreqs \<xi>"
+  "store (clear_locals \<xi>) = store \<xi>"
+  unfolding clear_locals_def by auto
+
+subsection "Auxilliary message handling definitions"
+
+definition is_newpkt
+where "is_newpkt \<xi> \<equiv> case msg \<xi> of
+                       Newpkt data' dip' \<Rightarrow> { \<xi>\<lparr>data := data', dip := dip'\<rparr> }
+                     | _ \<Rightarrow> {}"
+
+definition is_pkt
+where "is_pkt \<xi> \<equiv> case msg \<xi> of
+                    Pkt data' dip' oip' \<Rightarrow> { \<xi>\<lparr> data := data', dip := dip', oip := oip' \<rparr> }
+                  | _ \<Rightarrow> {}"
+
+definition is_rreq
+where "is_rreq \<xi> \<equiv> case msg \<xi> of
+                     Rreq hops' rreqid' dip' dsn' dsk' oip' osn' sip' \<Rightarrow>
+                       { \<xi>\<lparr> hops := hops', rreqid := rreqid', dip := dip', dsn := dsn',
+                            dsk := dsk', oip := oip', osn := osn', sip := sip' \<rparr> }
+                   | _ \<Rightarrow> {}"
+
+lemma is_rreq_asm [dest!]:
+  assumes "\<xi>' \<in> is_rreq \<xi>"
+    shows "(\<exists>hops' rreqid' dip' dsn' dsk' oip' osn' sip'.
+               msg \<xi> = Rreq hops' rreqid' dip' dsn' dsk' oip' osn' sip' \<and>
+               \<xi>' = \<xi>\<lparr> hops := hops', rreqid := rreqid', dip := dip', dsn := dsn',
+                       dsk := dsk', oip := oip', osn := osn', sip := sip' \<rparr>)"
+  using assms unfolding is_rreq_def
+  by (cases "msg \<xi>") simp_all
+
+definition is_rrep
+where "is_rrep \<xi> \<equiv> case msg \<xi> of
+                     Rrep hops' dip' dsn' oip' sip' \<Rightarrow>
+                       { \<xi>\<lparr> hops := hops', dip := dip', dsn := dsn', oip := oip', sip := sip' \<rparr> }
+                   | _ \<Rightarrow> {}"
+
+lemma is_rrep_asm [dest!]:
+  assumes "\<xi>' \<in> is_rrep \<xi>"
+    shows "(\<exists>hops' dip' dsn' oip' sip'.
+               msg \<xi> = Rrep hops' dip' dsn' oip' sip' \<and>
+               \<xi>' = \<xi>\<lparr> hops := hops', dip := dip', dsn := dsn', oip := oip', sip := sip' \<rparr>)"
+  using assms unfolding is_rrep_def
+  by (cases "msg \<xi>") simp_all
+
+definition is_rerr
+where "is_rerr \<xi> \<equiv> case msg \<xi> of
+                     Rerr dests' sip' \<Rightarrow> { \<xi>\<lparr> dests := dests', sip := sip' \<rparr> }
+                   | _ \<Rightarrow> {}"
+
+lemma is_rerr_asm [dest!]:
+  assumes "\<xi>' \<in> is_rerr \<xi>"
+    shows "(\<exists>dests' sip'.
+               msg \<xi> = Rerr dests' sip' \<and>
+               \<xi>' = \<xi>\<lparr> dests := dests', sip := sip' \<rparr>)"
+  using assms unfolding is_rerr_def
+  by (cases "msg \<xi>") simp_all
+
+lemmas is_msg_defs =
+  is_rerr_def is_rrep_def is_rreq_def is_pkt_def is_newpkt_def
+
+lemma is_msg_inv_ip [simp]:
+  "\<xi>' \<in> is_rerr \<xi>   \<Longrightarrow> ip \<xi>' = ip \<xi>"
+  "\<xi>' \<in> is_rrep \<xi>   \<Longrightarrow> ip \<xi>' = ip \<xi>"
+  "\<xi>' \<in> is_rreq \<xi>   \<Longrightarrow> ip \<xi>' = ip \<xi>"
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> ip \<xi>' = ip \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> ip \<xi>' = ip \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+lemma is_msg_inv_sn [simp]:
+  "\<xi>' \<in> is_rerr \<xi>   \<Longrightarrow> sn \<xi>' = sn \<xi>"
+  "\<xi>' \<in> is_rrep \<xi>   \<Longrightarrow> sn \<xi>' = sn \<xi>"
+  "\<xi>' \<in> is_rreq \<xi>   \<Longrightarrow> sn \<xi>' = sn \<xi>"
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> sn \<xi>' = sn \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> sn \<xi>' = sn \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+lemma is_msg_inv_rt [simp]:
+  "\<xi>' \<in> is_rerr \<xi>   \<Longrightarrow> rt \<xi>' = rt \<xi>"
+  "\<xi>' \<in> is_rrep \<xi>   \<Longrightarrow> rt \<xi>' = rt \<xi>"
+  "\<xi>' \<in> is_rreq \<xi>   \<Longrightarrow> rt \<xi>' = rt \<xi>"
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> rt \<xi>' = rt \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> rt \<xi>' = rt \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+lemma is_msg_inv_rreqs [simp]:
+  "\<xi>' \<in> is_rerr \<xi>   \<Longrightarrow> rreqs \<xi>' = rreqs \<xi>"
+  "\<xi>' \<in> is_rrep \<xi>   \<Longrightarrow> rreqs \<xi>' = rreqs \<xi>"
+  "\<xi>' \<in> is_rreq \<xi>   \<Longrightarrow> rreqs \<xi>' = rreqs \<xi>"
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> rreqs \<xi>' = rreqs \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> rreqs \<xi>' = rreqs \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+lemma is_msg_inv_store [simp]:
+  "\<xi>' \<in> is_rerr \<xi>   \<Longrightarrow> store \<xi>' = store \<xi>"
+  "\<xi>' \<in> is_rrep \<xi>   \<Longrightarrow> store \<xi>' = store \<xi>"
+  "\<xi>' \<in> is_rreq \<xi>   \<Longrightarrow> store \<xi>' = store \<xi>"
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> store \<xi>' = store \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> store \<xi>' = store \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+lemma is_msg_inv_sip [simp]:
+  "\<xi>' \<in> is_pkt \<xi>    \<Longrightarrow> sip \<xi>' = sip \<xi>"
+  "\<xi>' \<in> is_newpkt \<xi> \<Longrightarrow> sip \<xi>' = sip \<xi>"
+  unfolding is_msg_defs
+  by (cases "msg \<xi>", clarsimp+)+
+
+subsection "The protocol process"
+
+datatype pseqp =
+    PAodv
+  | PNewPkt
+  | PPkt
+  | PRreq
+  | PRrep
+  | PRerr
+
+fun nat_of_seqp :: "pseqp \<Rightarrow> nat"
+where
+  "nat_of_seqp PAodv  = 1"
+| "nat_of_seqp PPkt   = 2"
+| "nat_of_seqp PNewPkt   = 3"
+| "nat_of_seqp PRreq  = 4"
+| "nat_of_seqp PRrep  = 5"
+| "nat_of_seqp PRerr  = 6"
+
+instantiation "pseqp" :: ord
+begin
+definition less_eq_seqp [iff]: "l1 \<le> l2 = (nat_of_seqp l1 \<le> nat_of_seqp l2)"
+definition less_seqp [iff]:    "l1 < l2 = (nat_of_seqp l1 < nat_of_seqp l2)"
+instance ..
+end
+
+abbreviation AODV
+where
+  "AODV \<equiv> \<lambda>_. \<lbrakk>clear_locals\<rbrakk> call(PAodv)"
+
+abbreviation PKT
+where
+  "PKT args \<equiv>
+
+     \<lbrakk>\<xi>. let (data, dip, oip) = args \<xi> in
+         (clear_locals \<xi>) \<lparr> data := data, dip := dip, oip := oip \<rparr>\<rbrakk>
+     call(PPkt)"
+abbreviation NEWPKT
+where
+  "NEWPKT args \<equiv>
+     \<lbrakk>\<xi>. let (data, dip) = args \<xi> in
+         (clear_locals \<xi>) \<lparr> data := data, dip := dip \<rparr>\<rbrakk>
+     call(PNewPkt)"
+
+abbreviation RREQ
+where
+  "RREQ args \<equiv>
+     \<lbrakk>\<xi>. let (hops, rreqid, dip, dsn, dsk, oip, osn, sip) = args \<xi> in
+         (clear_locals \<xi>) \<lparr> hops := hops, rreqid := rreqid, dip := dip,
+                            dsn := dsn, dsk := dsk, oip := oip,
+                            osn := osn, sip := sip \<rparr>\<rbrakk>
+     call(PRreq)"
+
+abbreviation RREP
+where
+  "RREP args \<equiv>
+     \<lbrakk>\<xi>. let (hops, dip, dsn, oip, sip) = args \<xi> in
+         (clear_locals \<xi>) \<lparr> hops := hops, dip := dip, dsn := dsn,
+                            oip := oip, sip := sip \<rparr>\<rbrakk>
+     call(PRrep)"
+
+abbreviation RERR
+where
+  "RERR args \<equiv>
+     \<lbrakk>\<xi>. let (dests, sip) = args \<xi> in
+         (clear_locals \<xi>) \<lparr> dests := dests, sip := sip \<rparr>\<rbrakk>
+     call(PRerr)"
+
+fun \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V :: "(state, msg, pseqp, pseqp label) seqp_env"
+where
+  "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PAodv = labelled PAodv (
+     receive(\<lambda>msg' \<xi>. \<xi> \<lparr> msg := msg' \<rparr>).
+     (    \<langle>is_newpkt\<rangle> NEWPKT(\<lambda>\<xi>. (data \<xi>, ip \<xi>))
+       \<oplus> \<langle>is_pkt\<rangle> PKT(\<lambda>\<xi>. (data \<xi>, dip \<xi>, oip \<xi>))
+       \<oplus> \<langle>is_rreq\<rangle>
+            \<lbrakk>\<xi>. \<xi> \<lparr>rt := update (rt \<xi>) (sip \<xi>) (0, unk, val, 1, sip \<xi>, {}) \<rparr>\<rbrakk>
+            RREQ(\<lambda>\<xi>. (hops \<xi>, rreqid \<xi>, dip \<xi>, dsn \<xi>, dsk \<xi>, oip \<xi>, osn \<xi>, sip \<xi>))
+       \<oplus> \<langle>is_rrep\<rangle>
+            \<lbrakk>\<xi>. \<xi> \<lparr>rt := update (rt \<xi>) (sip \<xi>) (0, unk, val, 1, sip \<xi>, {}) \<rparr>\<rbrakk>
+            RREP(\<lambda>\<xi>. (hops \<xi>, dip \<xi>, dsn \<xi>, oip \<xi>, sip \<xi>))
+       \<oplus> \<langle>is_rerr\<rangle>
+            \<lbrakk>\<xi>. \<xi> \<lparr>rt := update (rt \<xi>) (sip \<xi>) (0, unk, val, 1, sip \<xi>, {}) \<rparr>\<rbrakk>
+            RERR(\<lambda>\<xi>. (dests \<xi>, sip \<xi>))
+     )
+     \<oplus> \<langle>\<lambda>\<xi>. { \<xi>\<lparr> dip := dip \<rparr> | dip. dip \<in> qD(store \<xi>) \<inter> vD(rt \<xi>) }\<rangle>
+          \<lbrakk>\<xi>. \<xi> \<lparr> data := hd(\<sigma>\<^bsub>queue\<^esub>(store \<xi>, dip \<xi>)) \<rparr>\<rbrakk>
+          unicast(\<lambda>\<xi>. the (nhop (rt \<xi>) (dip \<xi>)), \<lambda>\<xi>. pkt(data \<xi>, dip \<xi>, ip \<xi>)).
+            \<lbrakk>\<xi>. \<xi> \<lparr> store := the (drop (dip \<xi>) (store \<xi>)) \<rparr>\<rbrakk>
+            AODV()
+          \<triangleright> \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if (rip \<in> vD (rt \<xi>) \<and> nhop (rt \<xi>) rip = nhop (rt \<xi>) (dip \<xi>))
+                                     then Some (inc (sqn (rt \<xi>) rip)) else None) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                                     then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+             groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)). AODV()
+     \<oplus> \<langle>\<lambda>\<xi>. { \<xi>\<lparr> dip := dip \<rparr>
+             | dip. dip \<in> qD(store \<xi>) - vD(rt \<xi>) \<and> the (\<sigma>\<^bsub>p-flag\<^esub>(store \<xi>, dip)) = req }\<rangle>
+         \<lbrakk>\<xi>. \<xi> \<lparr> store := unsetRRF (store \<xi>) (dip \<xi>) \<rparr>\<rbrakk>
+         \<lbrakk>\<xi>. \<xi> \<lparr> sn := inc (sn \<xi>) \<rparr>\<rbrakk>
+         \<lbrakk>\<xi>. \<xi> \<lparr> rreqid := nrreqid (rreqs \<xi>) (ip \<xi>) \<rparr>\<rbrakk>
+         \<lbrakk>\<xi>. \<xi> \<lparr> rreqs := rreqs \<xi> \<union> {(ip \<xi>, rreqid \<xi>)} \<rparr>\<rbrakk>
+         broadcast(\<lambda>\<xi>. rreq(0, rreqid \<xi>, dip \<xi>, sqn (rt \<xi>) (dip \<xi>), sqnf (rt \<xi>) (dip \<xi>),
+                            ip \<xi>, sn \<xi>, ip \<xi>)). AODV())"
+
+|  "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PNewPkt = labelled PNewPkt (
+     \<langle>\<xi>. dip \<xi> = ip \<xi>\<rangle>
+        deliver(\<lambda>\<xi>. data \<xi>).AODV()
+     \<oplus> \<langle>\<xi>. dip \<xi> \<noteq> ip \<xi>\<rangle>
+        \<lbrakk>\<xi>. \<xi> \<lparr> store := add (data \<xi>) (dip \<xi>) (store \<xi>) \<rparr>\<rbrakk>
+        AODV())"
+
+| "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PPkt = labelled PPkt (
+     \<langle>\<xi>. dip \<xi> = ip \<xi>\<rangle>
+        deliver(\<lambda>\<xi>. data \<xi>).AODV()
+     \<oplus> \<langle>\<xi>. dip \<xi> \<noteq> ip \<xi>\<rangle>
+     (
+       \<langle>\<xi>. dip \<xi> \<in> vD (rt \<xi>)\<rangle>
+         unicast(\<lambda>\<xi>. the (nhop (rt \<xi>) (dip \<xi>)), \<lambda>\<xi>. pkt(data \<xi>, dip \<xi>, oip \<xi>)).AODV()
+         \<triangleright>
+           \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if (rip \<in> vD (rt \<xi>) \<and> nhop (rt \<xi>) rip = nhop (rt \<xi>) (dip \<xi>))
+                                   then Some (inc (sqn (rt \<xi>) rip)) else None) \<rparr>\<rbrakk>
+           \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+           \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+           \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+           \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                                   then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+           groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)).AODV()
+       \<oplus> \<langle>\<xi>. dip \<xi> \<notin> vD (rt \<xi>)\<rangle>
+       (
+           \<langle>\<xi>. dip \<xi> \<in> iD (rt \<xi>)\<rangle>
+             groupcast(\<lambda>\<xi>. the (precs (rt \<xi>) (dip \<xi>)),
+                       \<lambda>\<xi>. rerr([dip \<xi> \<mapsto> sqn (rt \<xi>) (dip \<xi>)], ip \<xi>)). AODV()
+           \<oplus> \<langle>\<xi>. dip \<xi> \<notin> iD (rt \<xi>)\<rangle>
+              AODV()
+       )
+     ))"
+
+| "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRreq = labelled PRreq (
+     \<langle>\<xi>. (oip \<xi>, rreqid \<xi>) \<in> rreqs \<xi>\<rangle>
+       AODV()
+     \<oplus> \<langle>\<xi>. (oip \<xi>, rreqid \<xi>) \<notin> rreqs \<xi>\<rangle>
+       \<lbrakk>\<xi>. \<xi> \<lparr> rt := update (rt \<xi>) (oip \<xi>) (osn \<xi>, kno, val, hops \<xi> + 1, sip \<xi>, {}) \<rparr>\<rbrakk>
+       \<lbrakk>\<xi>. \<xi> \<lparr> rreqs := rreqs \<xi> \<union> {(oip \<xi>, rreqid \<xi>)} \<rparr>\<rbrakk>
+       (
+         \<langle>\<xi>. dip \<xi> = ip \<xi>\<rangle>
+           \<lbrakk>\<xi>. \<xi> \<lparr> sn := max (sn \<xi>) (dsn \<xi>) \<rparr>\<rbrakk>
+           unicast(\<lambda>\<xi>. the (nhop (rt \<xi>) (oip \<xi>)), \<lambda>\<xi>. rrep(0, dip \<xi>, sn \<xi>, oip \<xi>, ip \<xi>)).AODV()
+           \<triangleright>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if (rip \<in> vD (rt \<xi>) \<and> nhop (rt \<xi>) rip = nhop (rt \<xi>) (oip \<xi>))
+                                     then Some (inc (sqn (rt \<xi>) rip)) else None) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                                     then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+             groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)).AODV()
+         \<oplus> \<langle>\<xi>. dip \<xi> \<noteq> ip \<xi>\<rangle>
+         (
+           \<langle>\<xi>. dip \<xi> \<in> vD (rt \<xi>) \<and> dsn \<xi> \<le> sqn (rt \<xi>) (dip \<xi>) \<and> sqnf (rt \<xi>) (dip \<xi>) = kno\<rangle>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := the (addpreRT (rt \<xi>) (dip \<xi>) {sip \<xi>}) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := the (addpreRT (rt \<xi>) (oip \<xi>) {the (nhop (rt \<xi>) (dip \<xi>))}) \<rparr>\<rbrakk> 
+             unicast(\<lambda>\<xi>. the (nhop (rt \<xi>) (oip \<xi>)), \<lambda>\<xi>. rrep(the (dhops (rt \<xi>) (dip \<xi>)), dip \<xi>,
+                         sqn (rt \<xi>) (dip \<xi>), oip \<xi>, ip \<xi>)).
+             AODV()
+           \<triangleright>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if (rip \<in> vD (rt \<xi>) \<and> nhop (rt \<xi>) rip = nhop (rt \<xi>) (oip \<xi>))
+                                     then Some (inc (sqn (rt \<xi>) rip)) else None) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                                     then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+             groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)).AODV()
+           \<oplus> \<langle>\<xi>. dip \<xi> \<notin> vD (rt \<xi>) \<or> sqn (rt \<xi>) (dip \<xi>) < dsn \<xi> \<or> sqnf (rt \<xi>) (dip \<xi>) = unk\<rangle>
+             broadcast(\<lambda>\<xi>. rreq(hops \<xi> + 1, rreqid \<xi>, dip \<xi>, max (sqn (rt \<xi>) (dip \<xi>)) (dsn \<xi>),
+                                dsk \<xi>, oip \<xi>, osn \<xi>, ip \<xi>)).
+             AODV()
+         )
+       ))"
+
+| "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRrep = labelled PRrep (
+       \<lbrakk>\<xi>. \<xi> \<lparr> rt := update (rt \<xi>) (dip \<xi>) (dsn \<xi>, kno, val, hops \<xi> + 1, sip \<xi>, {}) \<rparr> \<rbrakk>
+       (
+         \<langle>\<xi>. oip \<xi> = ip \<xi> \<rangle>
+            AODV()
+         \<oplus> \<langle>\<xi>. oip \<xi> \<noteq> ip \<xi> \<rangle>
+         (
+           \<langle>\<xi>. oip \<xi> \<in> vD (rt \<xi>) \<and> dip \<xi> \<in> vD (rt \<xi>)\<rangle>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := the (addpreRT (rt \<xi>) (dip \<xi>)
+                                               {the (nhop (rt \<xi>) (oip \<xi>))}) \<rparr>\<rbrakk> 
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := the (addpreRT (rt \<xi>) (the (nhop (rt \<xi>) (dip \<xi>))) {the (nhop (rt \<xi>) (oip \<xi>))}) \<rparr>\<rbrakk> 
+             unicast(\<lambda>\<xi>. the (nhop (rt \<xi>) (oip \<xi>)), \<lambda>\<xi>. rrep(the (dhops (rt \<xi>) (dip \<xi>)), dip \<xi>,
+                                                             sqn (rt \<xi>) (dip \<xi>), oip \<xi>, ip \<xi>)).
+             AODV()
+           \<triangleright>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if (rip \<in> vD (rt \<xi>) \<and> nhop (rt \<xi>) rip = nhop (rt \<xi>) (oip \<xi>))
+                                     then Some (inc (sqn (rt \<xi>) rip)) else None) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+             \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                                     then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+             groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)).AODV()
+           \<oplus> \<langle>\<xi>. oip \<xi> \<notin> vD (rt \<xi>) \<or>  dip \<xi> \<notin> vD (rt \<xi>)\<rangle>
+             AODV()
+         )
+       )
+     )"
+
+| "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRerr = labelled PRerr (
+     \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. case (dests \<xi>) rip of None \<Rightarrow> None
+                       | Some rsn \<Rightarrow> if rip \<in> vD (rt \<xi>) \<and> the (nhop (rt \<xi>) rip) = sip \<xi>
+                                       \<and> sqn (rt \<xi>) rip < rsn then Some rsn else None) \<rparr>\<rbrakk>
+     \<lbrakk>\<xi>. \<xi> \<lparr> rt := invalidate (rt \<xi>) (dests \<xi>) \<rparr>\<rbrakk>
+     \<lbrakk>\<xi>. \<xi> \<lparr> store := setRRF (store \<xi>) (dests \<xi>)\<rparr>\<rbrakk>
+     \<lbrakk>\<xi>. \<xi> \<lparr> pre := \<Union>{ the (precs (rt \<xi>) rip) | rip. rip \<in> dom (dests \<xi>) } \<rparr>\<rbrakk>
+     \<lbrakk>\<xi>. \<xi> \<lparr> dests := (\<lambda>rip. if ((dests \<xi>) rip \<noteq> None \<and> the (precs (rt \<xi>) rip) \<noteq> {})
+                             then (dests \<xi>) rip else None) \<rparr>\<rbrakk>
+     groupcast(\<lambda>\<xi>. pre \<xi>, \<lambda>\<xi>. rerr(dests \<xi>, ip \<xi>)). AODV())"
+
+declare \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V.simps [simp del, code del]
+lemmas \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_simps [simp, code] = \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V.simps [simplified]
+
+fun \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton
+where
+    "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PAodv   = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PAodv)"
+  | "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PNewPkt = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PNewPkt)"
+  | "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PPkt    = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PPkt)"
+  | "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PRreq   = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRreq)"
+  | "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PRrep   = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRrep)"
+  | "\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton PRerr   = seqp_skeleton (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRerr)"
+
+lemma \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton_wf [simp]:
+  "wellformed \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton"
+  proof (rule, intro allI)
+    fix pn pn'
+    show "call(pn') \<notin> stermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton pn)"
+      by (cases pn) simp_all
+  qed
+
+declare \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton.simps [simp del, code del]
+lemmas \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton_simps [simp, code]
+           = \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_skeleton.simps [simplified \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_simps seqp_skeleton.simps]
+
+lemma aodv_proc_cases [dest]:
+  fixes p pn
+  shows "p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pn) \<Longrightarrow>
+                                (p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PAodv) \<or> 
+                                 p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PNewPkt)  \<or>
+                                 p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PPkt)  \<or>
+                                 p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRreq) \<or>
+                                 p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRrep) \<or>
+                                 p \<in> ctermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PRerr))"
+  by (cases pn) simp_all
+
+definition \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V :: "ip \<Rightarrow> (state \<times> (state, msg, pseqp, pseqp label) seqp) set"
+where "\<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i \<equiv> {(aodv_init i, \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V PAodv)}"
+
+abbreviation paodv
+  :: "ip \<Rightarrow> (state \<times> (state, msg, pseqp, pseqp label) seqp, msg seq_action) automaton"
+where
+  "paodv i \<equiv> \<lparr> init = \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i, trans = seqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V \<rparr>"
+
+lemma aodv_trans: "trans (paodv i) = seqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V"
+  by simp
+
+lemma aodv_control_within [simp]: "control_within \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (init (paodv i))"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by (rule control_withinI) (auto simp del: \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_simps)
+
+lemma aodv_wf [simp]:
+  "wellformed \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V"
+  proof (rule, intro allI)
+    fix pn pn'
+    show "call(pn') \<notin> stermsl (\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pn)"
+      by (cases pn) simp_all
+  qed
+
+lemmas aodv_labels_not_empty [simp] = labels_not_empty [OF aodv_wf]
+
+lemma aodv_ex_label [intro]: "\<exists>l. l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p"
+  by (metis aodv_labels_not_empty all_not_in_conv)
+
+lemma aodv_ex_labelE [elim]:
+  assumes "\<forall>l\<in>labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p. P l p"
+      and "\<exists>p l. P l p \<Longrightarrow> Q"
+    shows "Q"
+  using assms by (metis aodv_ex_label)
+
+lemma aodv_simple_labels [simp]: "simple_labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V"
+  proof
+    fix pn p
+    assume "p\<in>subterms(\<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V pn)"
+    thus "\<exists>!l. labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p = {l}"
+    by (cases pn) (simp_all cong: seqp_congs | elim disjE)+
+  qed
+
+lemma \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_labels [simp]: "(\<xi>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i \<Longrightarrow>  labels \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V p = {PAodv-:0}"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by simp
+
+lemma aodv_init_kD_empty [simp]:
+  "(\<xi>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i \<Longrightarrow> kD (rt \<xi>) = {}"
+  unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def kD_def by simp
+
+lemma aodv_init_sip_not_ip [simp]: "\<not>(sip (aodv_init i) = i)" by simp
+
+lemma aodv_init_sip_not_ip' [simp]:
+  assumes "(\<xi>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i"
+    shows "sip \<xi> \<noteq> ip \<xi>"
+  using assms unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by simp
+
+lemma aodv_init_sip_not_i [simp]:
+  assumes "(\<xi>, p) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i"
+    shows "sip \<xi> \<noteq> i"
+  using assms unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by simp
+
+lemma clear_locals_sip_not_ip':
+  assumes "ip \<xi> = i"
+    shows "\<not>(sip (clear_locals \<xi>) = i)"
+  using assms by auto
+
+text \<open>Stop the simplifier from descending into process terms.\<close>
+declare seqp_congs [cong]
+
+text \<open>Configure the main invariant tactic for AODV.\<close>
+
+declare
+  \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_simps [cterms_env]
+  aodv_proc_cases [ctermsl_cases]
+  seq_invariant_ctermsI [OF aodv_wf aodv_control_within aodv_simple_labels aodv_trans,
+                            cterms_intros]
+  seq_step_invariant_ctermsI [OF aodv_wf aodv_control_within aodv_simple_labels aodv_trans,
+                                 cterms_intros]
+
+end

formal/afp/AODV/variants/b_fwdrreps/B_Aodv_Data.thy

@@ -0,0 +1,990 @@
+(*  Title:       variants/b_fwdrreps/Aodv_Data.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "Predicates and functions used in the AODV model"
+
+theory B_Aodv_Data
+imports B_Fwdrreps
+begin
+
+subsection "Sequence Numbers"
+
+text \<open>Sequence numbers approximate the relative freshness of routing information.\<close>
+
+definition inc :: "sqn \<Rightarrow> sqn"
+  where "inc sn \<equiv> if sn = 0 then sn else sn + 1"
+
+lemma less_than_inc [simp]: "x \<le> inc x"
+  unfolding inc_def by simp
+
+lemma inc_minus_suc_0 [simp]:
+  "inc x - Suc 0 = x"
+  unfolding inc_def by simp
+
+lemma inc_never_one' [simp, intro]: "inc x \<noteq> Suc 0"
+  unfolding inc_def by simp
+
+lemma inc_never_one [simp, intro]: "inc x \<noteq> 1"
+  by simp
+
+subsection "Modelling Routes"
+
+text \<open>
+  A route is a 6-tuple, @{term "(dsn, dsk, flag, hops, nhip, pre)"} where
+  @{term dsn} is the `destination sequence number', @{term dsk} is the
+  `destination-sequence-number status', @{term flag} is the route status,
+  @{term hops} is the number of hops to the destination, @{term nhip} is the
+  next hop toward the destination, and @{term pre} is the set of `precursor nodes'--those
+  interested in hearing about changes to the route.
+\<close>
+
+type_synonym r = "sqn \<times> k \<times> f \<times> nat \<times> ip \<times> ip set"
+
+definition proj2 :: "r \<Rightarrow> sqn" ("\<pi>\<^sub>2")
+  where "\<pi>\<^sub>2 \<equiv> \<lambda>(dsn, _, _, _, _, _). dsn"
+
+definition proj3 :: "r \<Rightarrow> k" ("\<pi>\<^sub>3")
+  where "\<pi>\<^sub>3 \<equiv> \<lambda>(_, dsk, _, _, _, _). dsk"
+
+definition proj4 :: "r \<Rightarrow> f" ("\<pi>\<^sub>4")
+  where "\<pi>\<^sub>4 \<equiv> \<lambda>(_, _, flag, _, _, _). flag"
+
+definition proj5 :: "r \<Rightarrow> nat" ("\<pi>\<^sub>5")
+  where "\<pi>\<^sub>5 \<equiv> \<lambda>(_, _, _, hops, _, _). hops"
+
+definition proj6 :: "r \<Rightarrow> ip" ("\<pi>\<^sub>6")
+  where "\<pi>\<^sub>6 \<equiv> \<lambda>(_, _, _, _, nhip, _). nhip"
+
+definition proj7 :: "r \<Rightarrow> ip set" ("\<pi>\<^sub>7")
+  where "\<pi>\<^sub>7 \<equiv> \<lambda>(_, _, _, _, _, pre). pre"
+
+lemma projs [simp]:
+  "\<pi>\<^sub>2(dsn, dsk, flag, hops, nhip, pre) = dsn"
+  "\<pi>\<^sub>3(dsn, dsk, flag, hops, nhip, pre) = dsk"
+  "\<pi>\<^sub>4(dsn, dsk, flag, hops, nhip, pre) = flag"
+  "\<pi>\<^sub>5(dsn, dsk, flag, hops, nhip, pre) = hops"
+  "\<pi>\<^sub>6(dsn, dsk, flag, hops, nhip, pre) = nhip"
+  "\<pi>\<^sub>7(dsn, dsk, flag, hops, nhip, pre) = pre"
+  by (clarsimp simp: proj2_def proj3_def proj4_def
+                     proj5_def proj6_def proj7_def)+
+
+lemma proj3_pred [intro]: "\<lbrakk> P kno; P unk \<rbrakk> \<Longrightarrow> P (\<pi>\<^sub>3 x)"
+  by (rule k.induct)
+
+lemma proj4_pred [intro]: "\<lbrakk> P val; P inv \<rbrakk> \<Longrightarrow> P (\<pi>\<^sub>4 x)"
+  by (rule f.induct)
+
+lemma proj6_pair_snd [simp]:
+  fixes dsn' r
+  shows "\<pi>\<^sub>6 (dsn', snd (r)) = \<pi>\<^sub>6(r)"
+  by (cases r) simp
+
+subsection "Routing Tables"
+
+text \<open>Routing tables map ip addresses to route entries.\<close>
+
+type_synonym rt = "ip \<rightharpoonup> r"
+
+syntax
+  "_Sigma_route" :: "rt \<Rightarrow> ip \<rightharpoonup> r"  ("\<sigma>\<^bsub>route\<^esub>'(_, _')")
+
+translations
+ "\<sigma>\<^bsub>route\<^esub>(rt, dip)" => "rt dip" 
+
+definition sqn :: "rt \<Rightarrow> ip \<Rightarrow> sqn"
+  where "sqn rt dip \<equiv> case \<sigma>\<^bsub>route\<^esub>(rt, dip) of Some r \<Rightarrow> \<pi>\<^sub>2(r) | None \<Rightarrow> 0"
+
+definition sqnf :: "rt \<Rightarrow> ip \<Rightarrow> k"
+  where "sqnf rt dip \<equiv> case \<sigma>\<^bsub>route\<^esub>(rt, dip) of Some r \<Rightarrow> \<pi>\<^sub>3(r) | None \<Rightarrow> unk"
+
+abbreviation flag :: "rt \<Rightarrow> ip \<rightharpoonup> f"
+  where "flag rt dip \<equiv> map_option \<pi>\<^sub>4 (\<sigma>\<^bsub>route\<^esub>(rt, dip))"
+
+abbreviation dhops :: "rt \<Rightarrow> ip \<rightharpoonup> nat"
+   where "dhops rt dip \<equiv> map_option \<pi>\<^sub>5 (\<sigma>\<^bsub>route\<^esub>(rt, dip))"
+
+abbreviation nhop :: "rt \<Rightarrow> ip \<rightharpoonup> ip"
+   where "nhop rt dip \<equiv> map_option \<pi>\<^sub>6 (\<sigma>\<^bsub>route\<^esub>(rt, dip))"
+
+abbreviation precs :: "rt \<Rightarrow> ip \<rightharpoonup> ip set"
+   where "precs rt dip \<equiv> map_option \<pi>\<^sub>7 (\<sigma>\<^bsub>route\<^esub>(rt, dip))"
+
+definition vD :: "rt \<Rightarrow> ip set"
+  where "vD rt \<equiv> {dip. flag rt dip = Some val}"
+
+definition iD :: "rt \<Rightarrow> ip set"
+  where "iD rt \<equiv> {dip. flag rt dip = Some inv}"
+
+definition kD :: "rt \<Rightarrow> ip set"
+  where "kD rt \<equiv> {dip. rt dip \<noteq> None}"
+
+lemma kD_is_vD_and_iD: "kD rt = vD rt \<union> iD rt"
+  unfolding kD_def vD_def iD_def by auto
+
+lemma vD_iD_gives_kD [simp]:
+   "\<And>ip rt. ip \<in> vD rt \<Longrightarrow> ip \<in> kD rt"
+   "\<And>ip rt. ip \<in> iD rt \<Longrightarrow> ip \<in> kD rt"
+  unfolding kD_is_vD_and_iD by simp_all
+
+lemma kD_Some [dest]:
+    fixes dip rt
+  assumes "dip \<in> kD rt"
+    shows "\<exists>dsn dsk flag hops nhip pre.
+           \<sigma>\<^bsub>route\<^esub>(rt, dip) = Some (dsn, dsk, flag, hops, nhip, pre)"
+  using assms unfolding kD_def by simp
+
+lemma kD_None [dest]:
+    fixes dip rt
+  assumes "dip \<notin> kD rt"
+    shows "\<sigma>\<^bsub>route\<^esub>(rt, dip) = None"
+  using assms unfolding kD_def
+  by (metis (mono_tags) mem_Collect_eq)
+
+lemma vD_Some [dest]:
+    fixes dip rt
+  assumes "dip \<in> vD rt"
+    shows "\<exists>dsn dsk hops nhip pre.
+           \<sigma>\<^bsub>route\<^esub>(rt, dip) = Some (dsn, dsk, val, hops, nhip, pre)"
+  using assms unfolding vD_def by simp
+
+lemma vD_empty [simp]: "vD Map.empty = {}"
+  unfolding vD_def by simp
+
+lemma iD_Some [dest]:
+    fixes dip rt
+  assumes "dip \<in> iD rt"
+    shows "\<exists>dsn dsk hops nhip pre.
+           \<sigma>\<^bsub>route\<^esub>(rt, dip) = Some (dsn, dsk, inv, hops, nhip, pre)"
+  using assms unfolding iD_def by simp
+
+lemma val_is_vD [elim]:
+    fixes ip rt
+  assumes "ip\<in>kD(rt)"
+      and "the (flag rt ip) = val"
+    shows "ip\<in>vD(rt)"
+  using assms unfolding vD_def by auto
+
+lemma inv_is_iD [elim]:
+    fixes ip rt
+  assumes "ip\<in>kD(rt)"
+      and "the (flag rt ip) = inv"
+    shows "ip\<in>iD(rt)"
+  using assms unfolding iD_def by auto
+
+lemma iD_flag_is_inv [elim, simp]:
+    fixes ip rt
+  assumes "ip\<in>iD(rt)"
+    shows "the (flag rt ip) = inv"
+  proof -
+    from \<open>ip\<in>iD(rt)\<close> have "ip\<in>kD(rt)" by auto
+    with assms show ?thesis unfolding iD_def by auto
+  qed
+
+lemma kD_but_not_vD_is_iD [elim]:
+    fixes ip rt
+  assumes "ip\<in>kD(rt)"
+      and "ip\<notin>vD(rt)"
+    shows "ip\<in>iD(rt)"
+  proof -
+    from \<open>ip\<in>kD(rt)\<close> obtain dsn dsk f hops nhop pre
+      where rtip: "rt ip = Some (dsn, dsk, f, hops, nhop, pre)"
+       by (metis kD_Some)
+    from \<open>ip\<notin>vD(rt)\<close> have "f \<noteq> val"
+    proof (rule contrapos_nn)
+      assume "f = val"
+      with rtip have "the (flag rt ip) = val" by simp
+      with \<open>ip\<in>kD(rt)\<close> show "ip\<in>vD(rt)" ..
+    qed
+    with rtip have "the (flag rt ip)= inv" by simp  
+    with \<open>ip\<in>kD(rt)\<close> show "ip\<in>iD(rt)" ..
+  qed
+
+lemma vD_or_iD [elim]:
+    fixes ip rt
+  assumes "ip\<in>kD(rt)"
+      and "ip\<in>vD(rt) \<Longrightarrow> P rt ip"
+      and "ip\<in>iD(rt) \<Longrightarrow> P rt ip"
+    shows "P rt ip"
+  proof -
+    from \<open>ip\<in>kD(rt)\<close> have "ip\<in>vD(rt) \<union> iD(rt)"
+      by (simp add: kD_is_vD_and_iD)
+    thus ?thesis by (auto elim: assms(2-3))
+  qed
+
+lemma proj5_eq_dhops: "\<And>dip rt. dip\<in>kD(rt) \<Longrightarrow> \<pi>\<^sub>5(the (rt dip)) = the (dhops rt dip)"
+  unfolding sqn_def by (drule kD_Some) clarsimp
+
+lemma proj4_eq_flag: "\<And>dip rt. dip\<in>kD(rt) \<Longrightarrow> \<pi>\<^sub>4(the (rt dip)) = the (flag rt dip)"
+  unfolding sqn_def by (drule kD_Some) clarsimp
+
+lemma proj2_eq_sqn: "\<And>dip rt. dip\<in>kD(rt) \<Longrightarrow> \<pi>\<^sub>2(the (rt dip)) = sqn rt dip"
+  unfolding sqn_def by (drule kD_Some) clarsimp
+
+lemma kD_sqnf_is_proj3 [simp]:
+  "\<And>ip rt. ip\<in>kD(rt) \<Longrightarrow> sqnf rt ip = \<pi>\<^sub>3(the (rt ip))"
+  unfolding sqnf_def by auto
+
+lemma vD_flag_val [simp]:
+  "\<And>dip rt. dip \<in> vD (rt) \<Longrightarrow> the (flag rt dip) = val"
+  unfolding vD_def by clarsimp
+
+lemma kD_update [simp]:
+  "\<And>rt nip v. kD (rt(nip \<mapsto> v)) = insert nip (kD rt)"
+  unfolding kD_def by auto
+
+lemma kD_empty [simp]: "kD Map.empty = {}"
+  unfolding kD_def by simp
+
+lemma ip_equal_or_known [elim]:
+  fixes rt ip ip'
+  assumes "ip = ip' \<or> ip\<in>kD(rt)"
+      and "ip = ip' \<Longrightarrow> P rt ip ip'"
+      and "\<lbrakk> ip \<noteq> ip'; ip\<in>kD(rt)\<rbrakk> \<Longrightarrow> P rt ip ip'"
+    shows "P rt ip ip'"
+  using assms by auto
+
+subsection "Updating Routing Tables"
+
+text \<open>Routing table entries are modified through explicit functions.
+      The properties of these functions are important in invariant proofs.\<close>
+
+subsubsection "Updating Precursor Lists"
+
+definition addpre :: "r \<Rightarrow> ip set \<Rightarrow> r"
+  where "addpre r npre \<equiv> let (dsn, dsk, flag, hops, nhip, pre) = r in
+                          (dsn, dsk, flag, hops, nhip, pre \<union> npre)"
+
+lemma proj2_addpre:
+  fixes v pre
+  shows "\<pi>\<^sub>2(addpre v pre) = \<pi>\<^sub>2(v)"
+  unfolding addpre_def by (cases v) simp
+
+lemma proj3_addpre:
+  fixes v pre
+  shows "\<pi>\<^sub>3(addpre v pre) = \<pi>\<^sub>3(v)"
+  unfolding addpre_def by (cases v) simp
+
+lemma proj4_addpre:
+  fixes v pre
+  shows "\<pi>\<^sub>4(addpre v pre) = \<pi>\<^sub>4(v)"
+  unfolding addpre_def by (cases v) simp
+
+lemma proj5_addpre:
+  fixes v pre
+  shows "\<pi>\<^sub>5(addpre v pre) = \<pi>\<^sub>5(v)"
+  unfolding addpre_def by (cases v) simp
+
+lemma proj6_addpre:
+  fixes dsn dsk flag hops nhip pre npre
+  shows "\<pi>\<^sub>6(addpre v npre) = \<pi>\<^sub>6(v)"
+  unfolding addpre_def by (cases v) simp
+
+lemma proj7_addpre:
+  fixes dsn dsk flag hops nhip pre npre
+  shows "\<pi>\<^sub>7(addpre v npre) = \<pi>\<^sub>7(v) \<union> npre"
+  unfolding addpre_def by (cases v) simp
+
+lemma addpre_empty: "addpre r {} = r"
+  unfolding addpre_def by simp
+
+lemma addpre_r:
+  "addpre (dsn, dsk, fl, hops, nhip, pre) npre = (dsn, dsk, fl, hops, nhip, pre \<union> npre)"
+  unfolding addpre_def by simp
+
+lemmas addpre_simps [simp] = proj2_addpre proj3_addpre proj4_addpre proj5_addpre
+                             proj6_addpre proj7_addpre addpre_empty addpre_r
+
+definition addpreRT :: "rt \<Rightarrow> ip \<Rightarrow> ip set \<rightharpoonup> rt"
+  where "addpreRT rt dip npre \<equiv>
+           map_option (\<lambda>s. rt (dip \<mapsto> addpre s npre)) (\<sigma>\<^bsub>route\<^esub>(rt, dip))"
+
+lemma snd_addpre [simp]:
+  "\<And>dsn dsn' v pre. (dsn, snd(addpre (dsn', v) pre)) = addpre (dsn, v) pre"
+  unfolding addpre_def by clarsimp
+
+lemma proj2_addpreRT [simp]:
+    fixes ip rt ip' npre
+  assumes "ip\<in>kD rt"
+      and "ip'\<in>kD rt"
+    shows "\<pi>\<^sub>2(the (the (addpreRT rt ip' npre) ip)) = \<pi>\<^sub>2(the (rt ip))"
+  using assms [THEN kD_Some] unfolding addpreRT_def by clarsimp
+
+lemma proj3_addpreRT [simp]:
+    fixes ip rt ip' npre
+  assumes "ip\<in>kD rt"
+      and "ip'\<in>kD rt"
+    shows "\<pi>\<^sub>3(the (the (addpreRT rt ip' npre) ip)) = \<pi>\<^sub>3(the (rt ip))"
+  using assms [THEN kD_Some] unfolding addpreRT_def by clarsimp
+
+lemma proj5_addpreRT [simp]:
+  "\<And>rt dip ip npre. dip\<in>kD(rt) \<Longrightarrow> \<pi>\<^sub>5(the (the (addpreRT rt dip npre) ip)) = \<pi>\<^sub>5(the (rt ip))"
+  unfolding addpreRT_def by auto
+
+lemma flag_addpreRT [simp]:
+    fixes rt pre ip dip
+  assumes "dip \<in> kD rt"
+    shows "flag (the (addpreRT rt dip pre)) ip = flag rt ip"
+  unfolding addpreRT_def
+  using assms [THEN kD_Some] by (clarsimp)
+
+  lemma kD_addpreRT [simp]:
+  fixes rt dip npre
+  assumes "dip \<in> kD rt"
+  shows "kD (the (addpreRT rt dip npre)) = kD rt"
+  unfolding kD_def addpreRT_def
+  using assms [THEN kD_Some]
+  by clarsimp blast
+
+lemma vD_addpreRT [simp]:
+  fixes rt dip npre
+  assumes "dip \<in> kD rt"
+  shows "vD (the (addpreRT rt dip npre)) = vD rt"
+  unfolding vD_def addpreRT_def
+  using assms [THEN kD_Some] by clarsimp auto
+
+lemma iD_addpreRT [simp]:
+  fixes rt dip npre
+  assumes "dip \<in> kD rt"
+  shows "iD (the (addpreRT rt dip npre)) = iD rt"
+  unfolding iD_def addpreRT_def
+  using assms [THEN kD_Some] by clarsimp auto
+
+lemma nhop_addpreRT [simp]:
+    fixes rt pre ip dip
+  assumes "dip \<in> kD rt"
+    shows "nhop (the (addpreRT rt dip pre)) ip = nhop rt ip"
+  unfolding sqn_def addpreRT_def
+  using assms [THEN kD_Some] by (clarsimp)
+
+lemma sqn_addpreRT [simp]:
+    fixes rt pre ip dip
+  assumes "dip \<in> kD rt"
+    shows "sqn (the (addpreRT rt dip pre)) ip = sqn rt ip"
+  unfolding sqn_def addpreRT_def
+  using assms [THEN kD_Some] by (clarsimp)
+
+lemma dhops_addpreRT [simp]:
+    fixes rt pre ip dip
+  assumes "dip \<in> kD rt"
+    shows "dhops (the (addpreRT rt dip pre)) ip = dhops rt ip"
+  unfolding addpreRT_def
+  using assms [THEN kD_Some] by (clarsimp)
+
+lemma sqnf_addpreRT [simp]:
+  "\<And>ip dip. ip\<in>kD(rt \<xi>) \<Longrightarrow> sqnf (the (addpreRT (rt \<xi>) ip npre)) dip = sqnf (rt \<xi>) dip"
+  unfolding sqnf_def addpreRT_def by auto
+
+subsubsection "Updating route entries"
+
+lemma in_kD_case [simp]:
+    fixes dip rt
+  assumes "dip \<in> kD(rt)"
+    shows "(case rt dip of None \<Rightarrow> en | Some r \<Rightarrow> es r) = es (the (rt dip))"
+  using assms [THEN kD_Some] by auto
+
+lemma not_in_kD_case [simp]:
+    fixes dip rt
+  assumes "dip \<notin> kD(rt)"
+    shows "(case rt dip of None \<Rightarrow> en | Some r \<Rightarrow> es r) = en"
+  using assms [THEN kD_None] by auto
+
+lemma rt_Some_sqn [dest]:
+    fixes rt and ip dsn dsk flag hops nhip pre
+  assumes "rt ip = Some (dsn, dsk, flag, hops, nhip, pre)"
+    shows "sqn rt ip = dsn"
+  unfolding sqn_def using assms by simp
+
+lemma not_kD_sqn [simp]:
+    fixes dip rt
+  assumes "dip \<notin> kD(rt)"
+    shows "sqn rt dip = 0"
+  using assms unfolding sqn_def
+  by simp
+
+definition update_arg_wf :: "r \<Rightarrow> bool"
+where "update_arg_wf r \<equiv> \<pi>\<^sub>4(r) = val \<and>
+                         (\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk) \<and>
+                         (\<pi>\<^sub>3(r) = unk \<longrightarrow> \<pi>\<^sub>5(r) = 1)"
+
+lemma update_arg_wf_gives_cases:
+  "\<And>r. update_arg_wf r \<Longrightarrow> (\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)"
+  unfolding update_arg_wf_def by simp
+
+lemma update_arg_wf_tuples [simp]:
+  "\<And>nhip pre. update_arg_wf (0, unk, val, Suc 0, nhip, pre)"
+  "\<And>n hops nhip pre. update_arg_wf (Suc n, kno, val, hops,  nhip, pre)"
+  unfolding update_arg_wf_def by auto
+
+lemma update_arg_wf_tuples' [elim]:
+  "\<And>n hops nhip pre. Suc 0 \<le> n \<Longrightarrow> update_arg_wf (n, kno, val, hops,  nhip, pre)"
+  unfolding update_arg_wf_def by auto
+
+lemma wf_r_cases [intro]:
+    fixes P r
+  assumes "update_arg_wf r"
+      and c1: "\<And>nhip pre. P (0, unk, val, Suc 0, nhip, pre)"
+      and c2: "\<And>dsn hops nhip pre. dsn > 0 \<Longrightarrow> P (dsn, kno, val, hops, nhip, pre)"
+    shows "P r"
+  proof -
+    obtain dsn dsk flag hops nhip pre
+    where *: "r = (dsn, dsk, flag, hops, nhip, pre)" by (cases r)
+    with \<open>update_arg_wf r\<close> have wf1: "flag = val"
+                            and wf2: "(dsn = 0) = (dsk = unk)"
+                            and wf3: "dsk = unk \<longrightarrow> (hops = 1)"
+      unfolding update_arg_wf_def by auto
+    have "P (dsn, dsk, flag, hops, nhip, pre)"
+    proof (cases dsk)
+      assume "dsk = unk"
+      moreover with wf2 wf3 have "dsn = 0" and "hops = Suc 0" by auto
+      ultimately show ?thesis using \<open>flag = val\<close> by simp (rule c1)
+    next
+      assume "dsk = kno"
+      moreover with wf2 have "dsn > 0" by simp
+      ultimately show ?thesis using \<open>flag = val\<close> by simp (rule c2)
+    qed
+    with * show "P r" by simp
+  qed
+
+definition update :: "rt \<Rightarrow> ip \<Rightarrow> r \<Rightarrow> rt"
+  where
+  "update rt ip r \<equiv>
+     case \<sigma>\<^bsub>route\<^esub>(rt, ip) of
+       None \<Rightarrow> rt (ip \<mapsto> r)
+     | Some s \<Rightarrow>
+          if \<pi>\<^sub>2(s) < \<pi>\<^sub>2(r) then rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(s)))
+          else if \<pi>\<^sub>2(s) = \<pi>\<^sub>2(r) \<and> (\<pi>\<^sub>5(s) > \<pi>\<^sub>5(r) \<or> \<pi>\<^sub>4(s) = inv)
+               then rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(s)))
+               else if \<pi>\<^sub>3(r) = unk
+                    then rt (ip \<mapsto> (\<pi>\<^sub>2(s), snd (addpre r (\<pi>\<^sub>7(s)))))
+                    else rt (ip \<mapsto> addpre s (\<pi>\<^sub>7(r)))"
+
+lemma update_simps [simp]:
+  fixes r s nrt nr nr' ns rt ip
+  defines "s \<equiv> the \<sigma>\<^bsub>route\<^esub>(rt, ip)"
+      and "nr \<equiv> addpre r (\<pi>\<^sub>7(s))"
+      and "nr' \<equiv> (\<pi>\<^sub>2(s), \<pi>\<^sub>3(nr), \<pi>\<^sub>4(nr), \<pi>\<^sub>5(nr), \<pi>\<^sub>6(nr), \<pi>\<^sub>7(nr))"
+      and "ns \<equiv> addpre s (\<pi>\<^sub>7(r))"
+  shows
+  "\<lbrakk>ip \<notin> kD(rt)\<rbrakk>                            \<Longrightarrow> update rt ip r = rt (ip \<mapsto> r)"
+  "\<lbrakk>ip \<in> kD(rt); sqn rt ip < \<pi>\<^sub>2(r)\<rbrakk>         \<Longrightarrow> update rt ip r = rt (ip \<mapsto> nr)"
+  "\<lbrakk>ip \<in> kD(rt); sqn rt ip = \<pi>\<^sub>2(r);
+                 the (dhops rt ip) > \<pi>\<^sub>5(r)\<rbrakk> \<Longrightarrow> update rt ip r = rt (ip \<mapsto> nr)"
+  "\<lbrakk>ip \<in> kD(rt); sqn rt ip = \<pi>\<^sub>2(r);
+                 flag rt ip = Some inv\<rbrakk>     \<Longrightarrow> update rt ip r = rt (ip \<mapsto> nr)"
+  "\<lbrakk>ip \<in> kD(rt); \<pi>\<^sub>3(r) = unk; (\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)\<rbrakk>  \<Longrightarrow> update rt ip r = rt (ip \<mapsto> nr')"
+  "\<lbrakk>ip \<in> kD(rt); sqn rt ip \<ge> \<pi>\<^sub>2(r); \<pi>\<^sub>3(r) = kno;
+    sqn rt ip = \<pi>\<^sub>2(r) \<Longrightarrow> the (dhops rt ip) \<le> \<pi>\<^sub>5(r) \<and> the (flag rt ip) = val \<rbrakk>
+                                            \<Longrightarrow> update rt ip r = rt (ip \<mapsto> ns)"
+  proof -
+    assume "ip\<notin>kD(rt)"
+    hence "\<sigma>\<^bsub>route\<^esub>(rt, ip) = None" ..
+    thus "update rt ip r = rt (ip \<mapsto> r)"
+      unfolding update_def by simp
+  next
+    assume "ip \<in> kD(rt)"
+       and "sqn rt ip < \<pi>\<^sub>2(r)"
+    from this(1) obtain dsn dsk fl hops nhip pre
+      where "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)
+    with \<open>sqn rt ip < \<pi>\<^sub>2(r)\<close> show "update rt ip r = rt (ip \<mapsto> nr)"
+      unfolding update_def nr_def s_def by auto
+  next
+    assume "ip \<in> kD(rt)"
+       and "sqn rt ip = \<pi>\<^sub>2(r)"
+       and "the (dhops rt ip) > \<pi>\<^sub>5(r)"
+    from this(1) obtain dsn dsk fl hops nhip pre
+      where "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)
+    with \<open>sqn rt ip = \<pi>\<^sub>2(r)\<close> and \<open>the (dhops rt ip) > \<pi>\<^sub>5(r)\<close>
+      show "update rt ip r = rt (ip \<mapsto> nr)"
+        unfolding update_def nr_def s_def by auto
+   next
+     assume "ip \<in> kD(rt)"
+        and "sqn rt ip = \<pi>\<^sub>2(r)"
+        and "flag rt ip = Some inv"
+    from this(1) obtain dsn dsk fl hops nhip pre
+      where "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)        
+     with \<open>sqn rt ip = \<pi>\<^sub>2(r)\<close> and \<open>flag rt ip = Some inv\<close>
+      show "update rt ip r = rt (ip \<mapsto> nr)"
+        unfolding update_def nr_def s_def by auto
+   next
+    assume "ip \<in> kD(rt)"
+       and "\<pi>\<^sub>3(r) = unk"
+       and "(\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)"
+    from this(1) obtain dsn dsk fl hops nhip pre
+      where "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)
+    with \<open>(\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)\<close> and \<open>\<pi>\<^sub>3(r) = unk\<close>
+      show "update rt ip r = rt (ip \<mapsto> nr')"
+        unfolding update_def nr'_def nr_def s_def
+      by (cases r) simp
+   next
+    assume "ip \<in> kD(rt)"
+       and otherassms: "sqn rt ip \<ge> \<pi>\<^sub>2(r)"
+           "\<pi>\<^sub>3(r) = kno"
+           "sqn rt ip = \<pi>\<^sub>2(r) \<Longrightarrow> the (dhops rt ip) \<le> \<pi>\<^sub>5(r) \<and> the (flag rt ip) = val"
+    from this(1) obtain dsn dsk fl hops nhip pre
+      where "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)
+     with otherassms show "update rt ip r = rt (ip \<mapsto> ns)"
+      unfolding update_def ns_def s_def by auto
+  qed
+
+lemma update_cases [elim]:
+  assumes "(\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)"
+      and c1: "\<lbrakk>ip \<notin> kD(rt)\<rbrakk> \<Longrightarrow> P (rt (ip \<mapsto> r))"
+
+      and c2: "\<lbrakk>ip \<in> kD(rt); sqn rt ip < \<pi>\<^sub>2(r)\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c3: "\<lbrakk>ip \<in> kD(rt); sqn rt ip = \<pi>\<^sub>2(r); the (dhops rt ip) > \<pi>\<^sub>5(r)\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c4: "\<lbrakk>ip \<in> kD(rt); sqn rt ip = \<pi>\<^sub>2(r); the (flag rt ip) = inv\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c5: "\<lbrakk>ip \<in> kD(rt); \<pi>\<^sub>3(r) = unk\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> (\<pi>\<^sub>2(the \<sigma>\<^bsub>route\<^esub>(rt, ip)), \<pi>\<^sub>3(r),
+                                  \<pi>\<^sub>4(r), \<pi>\<^sub>5(r), \<pi>\<^sub>6(r), \<pi>\<^sub>7(addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))))"
+      and c6: "\<lbrakk>ip \<in> kD(rt); sqn rt ip \<ge> \<pi>\<^sub>2(r); \<pi>\<^sub>3(r) = kno;
+                sqn rt ip = \<pi>\<^sub>2(r) \<Longrightarrow> the (dhops rt ip) \<le> \<pi>\<^sub>5(r) \<and> the (flag rt ip) = val\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre (the \<sigma>\<^bsub>route\<^esub>(rt, ip)) (\<pi>\<^sub>7(r))))"
+  shows "(P (update rt ip r))"
+  proof (cases "ip \<in> kD(rt)")
+    assume "ip \<notin> kD(rt)"
+    with c1 show ?thesis
+      by simp
+  next
+    assume "ip \<in> kD(rt)"
+    moreover then obtain dsn dsk fl hops nhip pre
+      where rteq: "rt ip = Some (dsn, dsk, fl, hops, nhip, pre)"
+        by (metis kD_Some)
+    moreover obtain dsn' dsk' fl' hops' nhip' pre'
+      where req: "r = (dsn', dsk', fl', hops', nhip', pre')"
+        by (cases r) metis
+    ultimately show ?thesis
+      using \<open>(\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)\<close>
+            c2 [OF \<open>ip\<in>kD(rt)\<close>]
+            c3 [OF \<open>ip\<in>kD(rt)\<close>]
+            c4 [OF \<open>ip\<in>kD(rt)\<close>]
+            c5 [OF \<open>ip\<in>kD(rt)\<close>]
+            c6 [OF \<open>ip\<in>kD(rt)\<close>]
+    unfolding update_def sqn_def by auto
+  qed
+
+lemma update_cases_kD:
+  assumes "(\<pi>\<^sub>2(r) = 0) = (\<pi>\<^sub>3(r) = unk)"
+      and "ip \<in> kD(rt)"
+      and c2: "sqn rt ip < \<pi>\<^sub>2(r) \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c3: "\<lbrakk>sqn rt ip = \<pi>\<^sub>2(r); the (dhops rt ip) > \<pi>\<^sub>5(r)\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c4: "\<lbrakk>sqn rt ip = \<pi>\<^sub>2(r); the (flag rt ip) = inv\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))"
+      and c5: "\<pi>\<^sub>3(r) = unk \<Longrightarrow> P (rt (ip \<mapsto> (\<pi>\<^sub>2(the \<sigma>\<^bsub>route\<^esub>(rt, ip)), \<pi>\<^sub>3(r),
+                                            \<pi>\<^sub>4(r), \<pi>\<^sub>5(r), \<pi>\<^sub>6(r),
+                                            \<pi>\<^sub>7(addpre r (\<pi>\<^sub>7(the \<sigma>\<^bsub>route\<^esub>(rt, ip)))))))"
+      and c6: "\<lbrakk>sqn rt ip \<ge> \<pi>\<^sub>2(r); \<pi>\<^sub>3(r) = kno;
+                sqn rt ip = \<pi>\<^sub>2(r) \<Longrightarrow> the (dhops rt ip) \<le> \<pi>\<^sub>5(r) \<and> the (flag rt ip) = val\<rbrakk>
+                \<Longrightarrow> P (rt (ip \<mapsto> addpre (the \<sigma>\<^bsub>route\<^esub>(rt, ip)) (\<pi>\<^sub>7(r))))"
+  shows "(P (update rt ip r))"
+  using assms(1) proof (rule update_cases)
+    assume "sqn rt ip < \<pi>\<^sub>2(r)"
+    thus "P (rt(ip \<mapsto> addpre r (\<pi>\<^sub>7(the (rt ip)))))" by (rule c2)
+  next
+    assume "sqn rt ip = \<pi>\<^sub>2(r)"
+       and "the (dhops rt ip) > \<pi>\<^sub>5(r)"
+    thus "P (rt(ip \<mapsto> addpre r (\<pi>\<^sub>7 (the (rt ip)))))"
+      by (rule c3)
+  next
+    assume "sqn rt ip = \<pi>\<^sub>2(r)"
+       and "the (flag rt ip) = inv"
+    thus "P (rt(ip \<mapsto> addpre r (\<pi>\<^sub>7 (the (rt ip)))))"
+      by (rule c4)
+  next
+    assume "\<pi>\<^sub>3(r) = unk"
+    thus "P (rt (ip \<mapsto> (\<pi>\<^sub>2(the \<sigma>\<^bsub>route\<^esub>(rt, ip)), \<pi>\<^sub>3(r), \<pi>\<^sub>4(r), \<pi>\<^sub>5(r), \<pi>\<^sub>6(r),
+                        \<pi>\<^sub>7(addpre r (\<pi>\<^sub>7(the (rt ip)))))))"
+      by (rule c5)
+  next
+    assume "sqn rt ip \<ge> \<pi>\<^sub>2(r)"
+       and "\<pi>\<^sub>3(r) = kno"
+       and "sqn rt ip = \<pi>\<^sub>2(r) \<Longrightarrow> the (dhops rt ip) \<le> \<pi>\<^sub>5(r) \<and> the (flag rt ip) = val"
+    thus "P (rt (ip \<mapsto> addpre (the (rt ip)) (\<pi>\<^sub>7(r))))"
+      by (rule c6)
+  qed (simp add: \<open>ip \<in> kD(rt)\<close>)
+
+lemma in_kD_after_update [simp]:
+  fixes rt nip dsn dsk flag hops nhip pre
+  shows "kD (update rt nip (dsn, dsk, flag, hops, nhip, pre)) = insert nip (kD rt)"
+  unfolding update_def
+  by (cases "rt nip") auto
+
+lemma nhop_of_update [simp]:
+  fixes rt dip dsn dsk flag hops nhip
+  assumes "rt \<noteq> update rt dip (dsn, dsk, flag, hops, nhip, {})"
+  shows "the (nhop (update rt dip (dsn, dsk, flag, hops, nhip, {})) dip) = nhip"
+  proof -
+  from assms
+  have update_neq: "\<And>v. rt dip = Some v \<Longrightarrow>
+          update rt dip (dsn, dsk, flag, hops, nhip, {})
+             \<noteq> rt(dip \<mapsto> addpre (the (rt dip)) (\<pi>\<^sub>7 (dsn, dsk, flag, hops, nhip, {})))"
+    by auto
+  show ?thesis
+    proof (cases "rt dip = None")
+      assume "rt dip = None"
+      thus "?thesis" unfolding update_def by clarsimp
+    next
+      assume "rt dip \<noteq> None"
+      then obtain v where "rt dip = Some v" by (metis not_None_eq)
+      with update_neq [OF this] show ?thesis
+        unfolding update_def by auto
+    qed
+  qed
+
+lemma sqn_if_updated:
+  fixes rip v rt ip
+  shows "sqn (\<lambda>x. if x = rip then Some v else rt x) ip
+         = (if ip = rip then \<pi>\<^sub>2(v) else sqn rt ip)"
+  unfolding sqn_def by simp
+
+lemma update_sqn [simp]:
+  fixes rt dip rip dsn dsk hops nhip pre
+  assumes "(dsn = 0) = (dsk = unk)"
+  shows "sqn rt dip \<le> sqn (update rt rip (dsn, dsk, val, hops, nhip, pre)) dip"
+  proof (rule update_cases)
+    show "(\<pi>\<^sub>2 (dsn, dsk, val, hops, nhip, pre) = 0) = (\<pi>\<^sub>3 (dsn, dsk, val, hops, nhip, pre) = unk)"
+      by simp (rule assms)
+  qed (clarsimp simp: sqn_if_updated sqn_def)+
+
+lemma sqn_update_bigger [simp]:
+    fixes rt ip ip' dsn dsk flag hops nhip pre
+  assumes "1 \<le> hops"
+    shows "sqn rt ip \<le> sqn (update rt ip' (dsn, dsk, flag, hops, nhip, pre)) ip"
+  using assms unfolding update_def sqn_def
+  by (clarsimp split: option.split) auto
+
+lemma dhops_update [intro]:
+    fixes rt dsn dsk flag hops ip rip nhip pre
+  assumes ex: "\<forall>ip\<in>kD rt. the (dhops rt ip) \<ge> 1"
+      and ip: "(ip = rip \<and> Suc 0 \<le> hops) \<or> (ip \<noteq> rip \<and> ip\<in>kD rt)"
+    shows "Suc 0 \<le> the (dhops (update rt rip (dsn, dsk, flag, hops, nhip, pre)) ip)"
+  using ip proof
+    assume "ip = rip \<and> Suc 0 \<le> hops" thus ?thesis
+      unfolding update_def using ex
+      by (cases "rip \<in> kD rt") (drule(1) bspec, auto)
+  next
+    assume "ip \<noteq> rip \<and> ip\<in>kD rt" thus ?thesis
+      using ex unfolding update_def
+      by (cases "rip\<in>kD rt") auto
+  qed
+
+lemma update_another [simp]:
+    fixes dip ip rt dsn dsk flag hops nhip pre
+  assumes "ip \<noteq> dip"
+    shows "(update rt dip (dsn, dsk, flag, hops, nhip, pre)) ip = rt ip"
+  using assms unfolding update_def
+  by (clarsimp split: option.split)
+
+lemma nhop_update_another [simp]:
+    fixes dip ip rt dsn dsk flag hops nhip pre
+  assumes "ip \<noteq> dip"
+    shows "nhop (update rt dip (dsn, dsk, flag, hops, nhip, pre)) ip = nhop rt ip"
+  using assms unfolding update_def
+  by (clarsimp split: option.split)
+
+lemma dhops_update_another [simp]:
+    fixes dip ip rt dsn dsk flag hops nhip pre
+  assumes "ip \<noteq> dip"
+    shows "dhops (update rt dip (dsn, dsk, flag, hops, nhip, pre)) ip = dhops rt ip"
+  using assms unfolding update_def
+  by (clarsimp split: option.split)
+
+lemma sqn_update_same [simp]:
+  "\<And>rt ip dsn dsk flag hops nhip pre. sqn (rt(ip \<mapsto> v)) ip = \<pi>\<^sub>2(v)"
+  unfolding sqn_def by simp
+
+lemma dhops_update_changed [simp]:
+    fixes rt dip osn hops nhip
+  assumes "rt \<noteq> update rt dip (osn, kno, val, hops, nhip, {})"
+    shows "the (dhops (update rt dip (osn, kno, val, hops, nhip, {})) dip) = hops"
+  using assms unfolding update_def                                                      
+  by (clarsimp split: option.split_asm option.split if_split_asm) auto
+
+lemma nhop_update_unk_val [simp]:
+  "\<And>rt dip ip dsn hops npre.
+   the (nhop (update rt dip (dsn, unk, val, hops, ip, npre)) dip) = ip"
+   unfolding update_def by (clarsimp split: option.split)
+
+lemma nhop_update_changed [simp]:
+    fixes rt dip dsn dsk flg hops sip
+  assumes "update rt dip (dsn, dsk, flg, hops, sip, {}) \<noteq> rt"
+    shows "the (nhop (update rt dip (dsn, dsk, flg, hops, sip, {})) dip) = sip"
+  using assms unfolding update_def
+  by (clarsimp split: option.splits if_split_asm) auto
+
+lemma update_rt_split_asm:
+  "\<And>rt ip dsn dsk flag hops sip.
+   P (update rt ip (dsn, dsk, flag, hops, sip, {}))
+   =
+   (\<not>(rt = update rt ip (dsn, dsk, flag, hops, sip, {}) \<and> \<not>P rt
+      \<or> rt \<noteq> update rt ip (dsn, dsk, flag, hops, sip, {})
+         \<and> \<not>P (update rt ip (dsn, dsk, flag, hops, sip, {}))))"
+  by auto
+
+lemma sqn_update [simp]: "\<And>rt dip dsn flg hops sip.
+  rt \<noteq> update rt dip (dsn, kno, flg, hops, sip, {})
+  \<Longrightarrow> sqn (update rt dip (dsn, kno, flg, hops, sip, {})) dip = dsn"
+  unfolding update_def by (clarsimp split: option.split if_split_asm) auto
+
+lemma sqnf_update [simp]: "\<And>rt dip dsn dsk flg hops sip.
+  rt \<noteq> update rt dip (dsn, dsk, flg, hops, sip, {})
+  \<Longrightarrow> sqnf (update rt dip (dsn, dsk, flg, hops, sip, {})) dip = dsk"
+  unfolding update_def sqnf_def
+  by (clarsimp split: option.splits if_split_asm) auto
+
+lemma update_kno_dsn_greater_zero:
+  "\<And>rt dip ip dsn hops npre. 1 \<le> dsn \<Longrightarrow> 1 \<le> (sqn (update rt dip (dsn, kno, val, hops, ip, npre)) dip)"
+   unfolding update_def 
+  by (clarsimp split: option.splits)
+
+lemma proj3_update [simp]: "\<And>rt dip dsn dsk flg hops sip.
+  rt \<noteq> update rt dip (dsn, dsk, flg, hops, sip, {})
+  \<Longrightarrow> \<pi>\<^sub>3(the (update rt dip (dsn, dsk, flg, hops, sip, {}) dip)) = dsk"
+  unfolding update_def sqnf_def
+  by (clarsimp split: option.splits if_split_asm) auto
+
+lemma nhop_update_changed_kno_val [simp]: "\<And>rt ip dsn dsk hops nhip.
+  rt \<noteq> update rt ip (dsn, kno, val, hops, nhip, {})
+   \<Longrightarrow> the (nhop (update rt ip (dsn, kno, val, hops, nhip, {})) ip) = nhip"
+  unfolding update_def
+  by (clarsimp split: option.split_asm option.split if_split_asm) auto
+
+lemma flag_update [simp]: "\<And>rt dip dsn flg hops sip.
+  rt \<noteq> update rt dip (dsn, kno, flg, hops, sip, {})
+  \<Longrightarrow> the (flag (update rt dip (dsn, kno, flg, hops, sip, {})) dip) = flg"
+  unfolding update_def
+  by (clarsimp split: option.split if_split_asm) auto
+
+lemma the_flag_Some [dest!]:
+    fixes ip rt
+  assumes "the (flag rt ip) = x"
+      and "ip \<in> kD rt"
+    shows "flag rt ip = Some x"
+  using assms by auto
+
+lemma kD_update_unchanged [dest]:
+    fixes rt dip dsn dsk flag hops nhip pre
+  assumes "rt = update rt dip (dsn, dsk, flag, hops, nhip, pre)"
+    shows "dip\<in>kD(rt)"
+  proof -
+    have "dip\<in>kD(update rt dip (dsn, dsk, flag, hops, nhip, pre))" by simp
+    with assms show ?thesis by simp
+  qed
+
+lemma nhop_update [simp]: "\<And>rt dip dsn dsk flg hops sip.
+  rt \<noteq> update rt dip (dsn, dsk, flg, hops, sip, {})
+  \<Longrightarrow> the (nhop (update rt dip (dsn, dsk, flg, hops, sip, {})) dip) = sip"
+  unfolding update_def sqnf_def
+  by (clarsimp split: option.splits if_split_asm) auto
+
+lemma sqn_update_another [simp]:
+    fixes dip ip rt dsn dsk flag hops nhip pre
+  assumes "ip \<noteq> dip"
+    shows "sqn (update rt dip (dsn, dsk, flag, hops, nhip, pre)) ip = sqn rt ip"
+  using assms unfolding update_def sqn_def
+  by (clarsimp split: option.splits) auto
+
+lemma sqnf_update_another [simp]:
+    fixes dip ip rt dsn dsk flag hops nhip pre
+  assumes "ip \<noteq> dip"
+    shows "sqnf (update rt dip (dsn, dsk, flag, hops, nhip, pre)) ip = sqnf rt ip"
+  using assms unfolding update_def sqnf_def
+  by (clarsimp split: option.splits) auto
+
+lemma vD_update_val [dest]:
+  "\<And>dip rt dip' dsn dsk hops nhip pre.
+   dip \<in> vD(update rt dip' (dsn, dsk, val, hops, nhip, pre)) \<Longrightarrow> (dip\<in>vD(rt) \<or> dip=dip')"
+   unfolding update_def vD_def by (clarsimp split: option.split_asm if_split_asm)
+
+subsubsection "Invalidating route entries"
+
+definition invalidate :: "rt \<Rightarrow> (ip \<rightharpoonup> sqn) \<Rightarrow> rt"
+where "invalidate rt dests \<equiv>
+  \<lambda>ip. case (rt ip, dests ip) of
+    (None, _) \<Rightarrow> None
+  | (Some s, None) \<Rightarrow> Some s
+  | (Some (_, dsk, _, hops, nhip, pre), Some rsn) \<Rightarrow>
+                      Some (rsn, dsk, inv, hops, nhip, pre)"
+
+lemma proj3_invalidate [simp]:
+  "\<And>dip. \<pi>\<^sub>3(the ((invalidate rt dests) dip)) = \<pi>\<^sub>3(the (rt dip))"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+lemma proj5_invalidate [simp]:
+  "\<And>dip. \<pi>\<^sub>5(the ((invalidate rt dests) dip)) = \<pi>\<^sub>5(the (rt dip))"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+lemma proj6_invalidate [simp]:
+  "\<And>dip. \<pi>\<^sub>6(the ((invalidate rt dests) dip)) = \<pi>\<^sub>6(the (rt dip))"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+lemma proj7_invalidate [simp]:
+  "\<And>dip. \<pi>\<^sub>7(the ((invalidate rt dests) dip)) = \<pi>\<^sub>7(the (rt dip))"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+lemma invalidate_kD_inv [simp]:
+  "\<And>rt dests. kD (invalidate rt dests) = kD rt"
+  unfolding invalidate_def kD_def
+  by (simp split: option.split)
+
+lemma invalidate_sqn:
+  fixes rt dip dests
+  assumes "\<forall>rsn. dests dip = Some rsn \<longrightarrow> sqn rt dip \<le> rsn"
+  shows "sqn rt dip \<le> sqn (invalidate rt dests) dip"
+  proof (cases "dip \<notin> kD(rt)")
+    assume "\<not> dip \<notin> kD(rt)"
+    hence "dip\<in>kD(rt)" by simp
+    then obtain dsn dsk flag hops nhip pre where "rt dip = Some (dsn, dsk, flag, hops, nhip, pre)"
+      by (metis kD_Some)
+    with assms show "sqn rt dip \<le> sqn (invalidate rt dests) dip"
+      by (cases "dests dip") (auto simp add: invalidate_def sqn_def)
+  qed simp
+
+lemma sqn_invalidate_in_dests [simp]:
+    fixes dests ipa rsn rt
+  assumes "dests ipa = Some rsn"
+      and "ipa\<in>kD(rt)"
+    shows "sqn (invalidate rt dests) ipa = rsn"
+  unfolding invalidate_def sqn_def
+  using assms(1) assms(2) [THEN kD_Some]
+  by clarsimp
+
+lemma dhops_invalidate [simp]:
+  "\<And>dip. the (dhops (invalidate rt dests) dip) = the (dhops rt dip)"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+lemma sqnf_invalidate [simp]:
+  "\<And>dip. sqnf (invalidate (rt \<xi>) (dests \<xi>)) dip = sqnf (rt \<xi>) dip"
+  unfolding sqnf_def invalidate_def by (clarsimp split: option.split)
+
+lemma nhop_invalidate [simp]:
+  "\<And>dip. the (nhop (invalidate (rt \<xi>) (dests \<xi>)) dip) = the (nhop (rt \<xi>) dip)"
+  unfolding invalidate_def by (clarsimp split: option.split)
+
+lemma invalidate_other [simp]:
+    fixes rt dests dip
+  assumes "dip\<notin>dom(dests)"
+    shows "invalidate rt dests dip = rt dip"
+  using assms unfolding invalidate_def
+  by (clarsimp split: option.split_asm)
+
+lemma invalidate_none [simp]:
+    fixes rt dests dip
+  assumes "dip\<notin>kD(rt)"
+    shows "invalidate rt dests dip = None"
+  using assms unfolding invalidate_def by clarsimp
+
+lemma vD_invalidate_vD_not_dests:
+  "\<And>dip rt dests. dip\<in>vD(invalidate rt dests) \<Longrightarrow> dip\<in>vD(rt) \<and> dests dip = None"
+  unfolding invalidate_def vD_def 
+  by (clarsimp split: option.split_asm)
+
+lemma sqn_invalidate_not_in_dests [simp]:
+  fixes dests dip rt
+  assumes "dip\<notin>dom(dests)"
+  shows "sqn (invalidate rt dests) dip = sqn rt dip"
+  using assms unfolding sqn_def by simp
+
+lemma invalidate_changes:
+    fixes rt dests dip dsn dsk flag hops nhip pre
+  assumes "invalidate rt dests dip = Some (dsn, dsk, flag, hops, nhip, pre)"
+    shows "  dsn = (case dests dip of None \<Rightarrow> \<pi>\<^sub>2(the (rt dip)) | Some rsn \<Rightarrow> rsn)
+           \<and> dsk = \<pi>\<^sub>3(the (rt dip))
+           \<and> flag = (if dests dip = None then \<pi>\<^sub>4(the (rt dip)) else inv)
+           \<and> hops = \<pi>\<^sub>5(the (rt dip))
+           \<and> nhip = \<pi>\<^sub>6(the (rt dip))
+           \<and> pre = \<pi>\<^sub>7(the (rt dip))"
+  using assms unfolding invalidate_def
+  by (cases "rt dip", clarsimp, cases "dests dip") auto
+
+
+lemma proj3_inv: "\<And>dip rt dests. dip\<in>kD (rt)
+                      \<Longrightarrow> \<pi>\<^sub>3(the (invalidate rt dests dip)) = \<pi>\<^sub>3(the (rt dip))"
+  by (clarsimp simp: invalidate_def kD_def split: option.split)
+
+lemma dests_iD_invalidate [simp]:
+  assumes "dests ip = Some rsn"
+      and "ip\<in>kD(rt)"
+    shows "ip\<in>iD(invalidate rt dests)"
+  using assms(1) assms(2) [THEN kD_Some] unfolding invalidate_def iD_def
+  by (clarsimp split: option.split)
+
+subsection "Route Requests"
+
+text \<open>Generate a fresh route request identifier.\<close>
+
+definition nrreqid :: "(ip \<times> rreqid) set \<Rightarrow> ip \<Rightarrow> rreqid"
+  where "nrreqid rreqs ip \<equiv> Max ({n. (ip, n) \<in> rreqs} \<union> {0}) + 1"
+
+subsection "Queued Packets"
+
+text \<open>Functions for sending data packets.\<close>
+
+type_synonym store = "ip \<rightharpoonup> (p \<times> data list)"
+
+definition sigma_queue :: "store \<Rightarrow> ip \<Rightarrow> data list"    ("\<sigma>\<^bsub>queue\<^esub>'(_, _')")
+  where "\<sigma>\<^bsub>queue\<^esub>(store, dip) \<equiv> case store dip of None \<Rightarrow> [] | Some (p, q) \<Rightarrow> q"
+
+definition qD :: "store \<Rightarrow> ip set"
+  where "qD \<equiv> dom"
+
+definition add :: "data \<Rightarrow> ip \<Rightarrow> store \<Rightarrow> store"
+  where "add d dip store \<equiv> case store dip of
+                              None \<Rightarrow> store (dip \<mapsto> (req, [d]))
+                            | Some (p, q) \<Rightarrow> store (dip \<mapsto> (p, q @ [d]))"
+
+lemma qD_add [simp]:
+  fixes d dip store
+  shows "qD(add d dip store) = insert dip (qD store)"
+  unfolding add_def Let_def qD_def
+  by (clarsimp split: option.split)
+
+definition drop :: "ip \<Rightarrow> store \<rightharpoonup> store"
+  where "drop dip store \<equiv>
+    map_option (\<lambda>(p, q). if tl q = [] then store (dip := None)
+                                      else store (dip \<mapsto> (p, tl q))) (store dip)"
+
+definition sigma_p_flag :: "store \<Rightarrow> ip \<rightharpoonup> p" ("\<sigma>\<^bsub>p-flag\<^esub>'(_, _')")
+  where "\<sigma>\<^bsub>p-flag\<^esub>(store, dip) \<equiv> map_option fst (store dip)"
+
+definition unsetRRF :: "store \<Rightarrow> ip \<Rightarrow> store"
+  where "unsetRRF store dip \<equiv> case store dip of
+                                None \<Rightarrow> store
+                              | Some (p, q) \<Rightarrow> store (dip \<mapsto> (noreq, q))"
+
+definition setRRF :: "store \<Rightarrow> (ip \<rightharpoonup> sqn) \<Rightarrow> store"
+  where "setRRF store dests \<equiv> \<lambda>dip. if dests dip = None then store dip
+                                    else map_option (\<lambda>(_, q). (req, q)) (store dip)"
+
+subsection "Comparison with the original technical report"
+
+text \<open>
+  The major differences with the AODV technical report of Fehnker et al are:
+  \begin{enumerate}
+  \item @{term nhop} is partial, thus a `@{term the}' is needed, similarly for @{term dhops}
+        and @{term addpreRT}.
+  \item @{term precs} is partial.
+  \item @{term "\<sigma>\<^bsub>p-flag\<^esub>(store, dip)"} is partial.
+  \item The routing table (@{typ rt}) is modelled as a map (@{typ "ip \<Rightarrow> r option"})
+        rather than a set of 7-tuples, likewise, the @{typ r} is a 6-tuple rather than
+        a 7-tuple, i.e., the destination ip-address (@{term "dip"}) is taken from the
+        argument to the function, rather than a part of the result. Well-definedness then
+        follows from the structure of the type and more related facts are available
+        automatically, rather than having to be acquired through tedious proofs.
+  \item Similar remarks hold for the dests mapping passed to @{term "invalidate"},
+        and @{term "store"}.
+  \end{enumerate}
+\<close>
+
+end
+

formal/afp/AODV/variants/b_fwdrreps/B_Aodv_Loop_Freedom.thy

@@ -0,0 +1,369 @@
+(*  Title:       variants/b_fwdrreps/Aodv_Loop_Freedom.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "Lift and transfer invariants to show loop freedom"
+
+theory B_Aodv_Loop_Freedom
+imports AWN.OClosed_Transfer AWN.Qmsg_Lifting B_Global_Invariants B_Loop_Freedom
+begin
+
+subsection \<open>Lift to parallel processes with queues\<close>
+
+lemma par_step_no_change_on_send_or_receive:
+    fixes \<sigma> s a \<sigma>' s'
+  assumes "((\<sigma>, s), a, (\<sigma>', s')) \<in> oparp_sos i (oseqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i) (seqp_sos \<Gamma>\<^sub>Q\<^sub>M\<^sub>S\<^sub>G)"
+      and "a \<noteq> \<tau>"
+    shows "\<sigma>' i = \<sigma> i"
+  using assms by (rule qmsg_no_change_on_send_or_receive)
+
+lemma par_nhop_quality_increases:
+  shows "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile> (otherwith ((=)) {i} (orecvmsg (\<lambda>\<sigma> m.
+                                    msg_fresh \<sigma> m \<and> msg_zhops m)),
+                                  other quality_increases {i} \<rightarrow>)
+                        global (\<lambda>\<sigma>. \<forall>dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                          in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                                             \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+    proof (rule lift_into_qmsg [OF seq_nhop_quality_increases])
+    show "opaodv i \<Turnstile>\<^sub>A (otherwith ((=)) {i}
+                         (orecvmsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)),
+                        other quality_increases {i} \<rightarrow>)
+                       globala (\<lambda>(\<sigma>, _, \<sigma>'). quality_increases (\<sigma> i) (\<sigma>' i))"
+    proof (rule ostep_invariant_weakenE [OF oquality_increases], simp_all)
+      fix t :: "(((nat \<Rightarrow> state) \<times> (state, msg, pseqp, pseqp label) seqp), msg seq_action) transition"
+      assume "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<sigma>, _), _, (\<sigma>', _)). \<forall>j. quality_increases (\<sigma> j) (\<sigma>' j)) t"
+      thus "quality_increases (fst (fst t) i) (fst (snd (snd t)) i)"
+        by (cases t) (clarsimp dest!: onllD, metis aodv_ex_label)
+    next
+      fix \<sigma> \<sigma>' a
+      assume "otherwith ((=)) {i}
+                (orecvmsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)) \<sigma> \<sigma>' a"
+      thus "otherwith quality_increases {i} (orecvmsg (\<lambda>_. rreq_rrep_sn)) \<sigma> \<sigma>' a"
+        by - (erule weaken_otherwith, auto)
+    qed
+  qed auto
+
+lemma par_rreq_rrep_sn_quality_increases:
+  "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                            globala (\<lambda>(\<sigma>, _, \<sigma>'). quality_increases (\<sigma> i) (\<sigma>' i))"
+  proof -
+    have "opaodv i \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                globala (\<lambda>(\<sigma>, _, \<sigma>'). quality_increases (\<sigma> i) (\<sigma>' i))"
+      by (rule ostep_invariant_weakenE [OF olocal_quality_increases])
+         (auto dest!: onllD seqllD elim!: aodv_ex_labelE)
+    hence "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                                   globala (\<lambda>(\<sigma>, _, \<sigma>'). quality_increases (\<sigma> i) (\<sigma>' i))"
+      by (rule lift_step_into_qmsg_statelessassm) simp_all
+    thus ?thesis by rule auto
+  qed
+
+lemma par_rreq_rrep_nsqn_fresh_any_step:
+  shows "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>,
+                                   other (\<lambda>_ _. True) {i} \<rightarrow>)
+                                  globala (\<lambda>(\<sigma>, a, \<sigma>'). anycast (msg_fresh \<sigma>) a)"
+  proof -
+    have "opaodv i \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. (orecvmsg (\<lambda>_. rreq_rrep_sn)) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                       globala (\<lambda>(\<sigma>, a, \<sigma>'). anycast (msg_fresh \<sigma>) a)"
+    proof (rule ostep_invariant_weakenE [OF rreq_rrep_nsqn_fresh_any_step_invariant])
+      fix t
+      assume "onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>((\<sigma>, _), a, _). anycast (msg_fresh \<sigma>) a) t"
+      thus "globala (\<lambda>(\<sigma>, a, \<sigma>'). anycast (msg_fresh \<sigma>) a) t"
+        by (cases t) (clarsimp dest!: onllD, metis aodv_ex_label)
+    qed auto
+    hence "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. (orecvmsg (\<lambda>_. rreq_rrep_sn)) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                                    globala (\<lambda>(\<sigma>, a, \<sigma>'). anycast (msg_fresh \<sigma>) a)"
+      by (rule lift_step_into_qmsg_statelessassm) simp_all
+    thus ?thesis by rule auto
+  qed
+
+lemma par_anycast_msg_zhops:
+  shows "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                                  globala (\<lambda>(_, a, _). anycast msg_zhops a)"
+  proof -
+    from anycast_msg_zhops initiali_aodv oaodv_trans aodv_trans
+      have "opaodv i \<Turnstile>\<^sub>A (act TT, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                         seqll i (onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(_, a, _). anycast msg_zhops a))"
+        by (rule open_seq_step_invariant)
+    hence "opaodv i \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                            globala (\<lambda>(_, a, _). anycast msg_zhops a)"
+    proof (rule ostep_invariant_weakenE)
+      fix t :: "(((nat \<Rightarrow> state) \<times> (state, msg, pseqp, pseqp label) seqp), msg seq_action) transition"
+      assume "seqll i (onll \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V (\<lambda>(_, a, _). anycast msg_zhops a)) t"
+      thus "globala (\<lambda>(_, a, _). anycast msg_zhops a) t"
+        by (cases t) (clarsimp dest!: seqllD onllD, metis aodv_ex_label)
+    qed simp_all
+    hence "opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. orecvmsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+                                     globala (\<lambda>(_, a, _). anycast msg_zhops a)"
+      by (rule lift_step_into_qmsg_statelessassm) simp_all
+    thus ?thesis by rule auto
+  qed
+
+subsection \<open>Lift to nodes\<close>
+
+lemma node_step_no_change_on_send_or_receive:
+  assumes "((\<sigma>, NodeS i P R), a, (\<sigma>', NodeS i' P' R')) \<in> onode_sos
+                                      (oparp_sos i (oseqp_sos \<Gamma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i) (seqp_sos \<Gamma>\<^sub>Q\<^sub>M\<^sub>S\<^sub>G))"
+      and "a \<noteq> \<tau>"
+    shows "\<sigma>' i = \<sigma> i"
+  using assms
+  by (cases a) (auto elim!: par_step_no_change_on_send_or_receive)
+
+lemma node_nhop_quality_increases:
+  shows "\<langle> i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R \<rangle>\<^sub>o \<Turnstile>
+           (otherwith ((=)) {i}
+              (oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)),
+              other quality_increases {i}
+            \<rightarrow>) global (\<lambda>\<sigma>. \<forall>dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                  in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                                     \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+  by (rule node_lift [OF par_nhop_quality_increases]) auto
+
+lemma node_quality_increases:
+  "\<langle> i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R \<rangle>\<^sub>o \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma>,
+                                         other (\<lambda>_ _. True) {i} \<rightarrow>)
+                            globala (\<lambda>(\<sigma>, _, \<sigma>'). quality_increases (\<sigma> i) (\<sigma>' i))"
+  by (rule node_lift_step_statelessassm [OF par_rreq_rrep_sn_quality_increases]) simp
+
+lemma node_rreq_rrep_nsqn_fresh_any_step:
+  shows "\<langle> i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R \<rangle>\<^sub>o \<Turnstile>\<^sub>A
+          (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+          globala (\<lambda>(\<sigma>, a, \<sigma>'). castmsg (msg_fresh \<sigma>) a)"
+  by (rule node_lift_anycast_statelessassm [OF par_rreq_rrep_nsqn_fresh_any_step])
+
+lemma node_anycast_msg_zhops:
+  shows "\<langle> i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R \<rangle>\<^sub>o \<Turnstile>\<^sub>A
+          (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma>, other (\<lambda>_ _. True) {i} \<rightarrow>)
+          globala (\<lambda>(_, a, _). castmsg msg_zhops a)"
+  by (rule node_lift_anycast_statelessassm [OF par_anycast_msg_zhops])
+
+lemma node_silent_change_only:
+  shows "\<langle> i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<^sub>i \<rangle>\<^sub>o \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_ _. True) \<sigma>,
+                                               other (\<lambda>_ _. True) {i} \<rightarrow>)
+          globala (\<lambda>(\<sigma>, a, \<sigma>'). a \<noteq> \<tau> \<longrightarrow> \<sigma>' i = \<sigma> i)"
+  proof (rule ostep_invariantI, simp (no_asm), rule impI)
+    fix \<sigma> \<zeta> a \<sigma>' \<zeta>'
+    assume or: "(\<sigma>, \<zeta>) \<in> oreachable (\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<^sub>i\<rangle>\<^sub>o)
+                                    (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_ _. True) \<sigma>)
+                                    (other (\<lambda>_ _. True) {i})"
+      and tr: "((\<sigma>, \<zeta>), a, (\<sigma>', \<zeta>')) \<in> trans (\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<^sub>i\<rangle>\<^sub>o)"
+      and "a \<noteq> \<tau>\<^sub>n"
+    from or obtain p R where "\<zeta> = NodeS i p R"
+      by - (drule node_net_state, metis)
+    with tr have "((\<sigma>, NodeS i p R), a, (\<sigma>', \<zeta>'))
+                     \<in> onode_sos (oparp_sos i (trans (opaodv i)) (trans qmsg))"
+      by simp
+    thus "\<sigma>' i = \<sigma> i" using \<open>a \<noteq> \<tau>\<^sub>n\<close> 
+      by (cases rule: onode_sos.cases)
+         (auto elim: qmsg_no_change_on_send_or_receive)
+  qed
+
+subsection \<open>Lift to partial networks\<close>
+
+lemma arrive_rreq_rrep_nsqn_fresh_inc_sn [simp]:
+  assumes "oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> P \<sigma> m) \<sigma> m"
+    shows "oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma> m"
+  using assms by (cases m) auto
+
+lemma opnet_nhop_quality_increases:
+  shows "opnet (\<lambda>i. opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg) p \<Turnstile>
+           (otherwith ((=)) (net_tree_ips p)
+              (oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)),
+               other quality_increases (net_tree_ips p) \<rightarrow>)
+              global (\<lambda>\<sigma>. \<forall>i\<in>net_tree_ips p. \<forall>dip.
+                          let nhip = the (nhop (rt (\<sigma> i)) dip)
+                          in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                             \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+  proof (rule pnet_lift [OF node_nhop_quality_increases])
+    fix i R
+    have "\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o \<Turnstile>\<^sub>A (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma>,
+                                              other (\<lambda>_ _. True) {i} \<rightarrow>) globala (\<lambda>(\<sigma>, a, \<sigma>').
+            castmsg (\<lambda>m. msg_fresh \<sigma> m \<and> msg_zhops m) a)"
+    proof (rule ostep_invariantI, simp (no_asm))
+      fix \<sigma> s a \<sigma>' s'
+      assume or: "(\<sigma>, s) \<in> oreachable (\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o)
+                                      (\<lambda>\<sigma> _. oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma>)
+                                      (other (\<lambda>_ _. True) {i})"
+         and tr: "((\<sigma>, s), a, (\<sigma>', s')) \<in> trans (\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o)"
+         and am: "oarrivemsg (\<lambda>_. rreq_rrep_sn) \<sigma> a"
+      from or tr am have "castmsg (msg_fresh \<sigma>) a"
+        by (auto dest!: ostep_invariantD [OF node_rreq_rrep_nsqn_fresh_any_step])
+      moreover from or tr am have "castmsg (msg_zhops) a"
+        by (auto dest!: ostep_invariantD [OF node_anycast_msg_zhops])
+      ultimately show "castmsg (\<lambda>m. msg_fresh \<sigma> m \<and> msg_zhops m) a"
+        by (case_tac a) auto
+    qed
+    thus "\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o \<Turnstile>\<^sub>A
+            (\<lambda>\<sigma> _. oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m) \<sigma>,
+             other quality_increases {i} \<rightarrow>) globala (\<lambda>(\<sigma>, a, _).
+               castmsg (\<lambda>m. msg_fresh \<sigma> m \<and> msg_zhops m) a)"
+      by rule auto
+  next
+    fix i R
+    show "\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o \<Turnstile>\<^sub>A
+            (\<lambda>\<sigma> _. oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m) \<sigma>,
+             other quality_increases {i} \<rightarrow>) globala (\<lambda>(\<sigma>, a, \<sigma>').
+               a \<noteq> \<tau> \<and> (\<forall>d. a \<noteq> i:deliver(d)) \<longrightarrow> \<sigma> i = \<sigma>' i)"
+      by (rule ostep_invariant_weakenE [OF node_silent_change_only]) auto
+  next
+    fix i R
+    show "\<langle>i : opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg : R\<rangle>\<^sub>o \<Turnstile>\<^sub>A
+            (\<lambda>\<sigma> _. oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m) \<sigma>,
+             other quality_increases {i} \<rightarrow>) globala (\<lambda>(\<sigma>, a, \<sigma>').
+               a = \<tau> \<or> (\<exists>d. a = i:deliver(d)) \<longrightarrow> quality_increases (\<sigma> i) (\<sigma>' i))"
+      by (rule ostep_invariant_weakenE [OF node_quality_increases]) auto
+  qed simp_all
+
+subsection \<open>Lift to closed networks\<close>
+
+lemma onet_nhop_quality_increases:
+  shows "oclosed (opnet (\<lambda>i. opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg) p)
+           \<Turnstile> (\<lambda>_ _ _. True, other quality_increases (net_tree_ips p) \<rightarrow>)
+              global (\<lambda>\<sigma>. \<forall>i\<in>net_tree_ips p. \<forall>dip.
+                          let nhip = the (nhop (rt (\<sigma> i)) dip)
+                          in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                             \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+  (is "_ \<Turnstile> (_, ?U \<rightarrow>) ?inv")
+  proof (rule inclosed_closed)
+    from opnet_nhop_quality_increases
+      show "opnet (\<lambda>i. opaodv i \<langle>\<langle>\<^bsub>i\<^esub> qmsg) p
+               \<Turnstile> (otherwith ((=)) (net_tree_ips p) inoclosed, ?U \<rightarrow>) ?inv"
+    proof (rule oinvariant_weakenE)
+      fix \<sigma> \<sigma>' :: "ip \<Rightarrow> state" and a :: "msg node_action"
+      assume "otherwith ((=)) (net_tree_ips p) inoclosed \<sigma> \<sigma>' a"
+      thus "otherwith ((=)) (net_tree_ips p)
+              (oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m)) \<sigma> \<sigma>' a"
+      proof (rule otherwithEI)
+        fix \<sigma> :: "ip \<Rightarrow> state" and a :: "msg node_action"
+        assume "inoclosed \<sigma> a"
+        thus "oarrivemsg (\<lambda>\<sigma> m. msg_fresh \<sigma> m \<and> msg_zhops m) \<sigma> a"
+        proof (cases a)
+          fix ii ni ms
+          assume "a = ii\<not>ni:arrive(ms)"
+          moreover with \<open>inoclosed \<sigma> a\<close> obtain d di where "ms = newpkt(d, di)"
+            by (cases ms) auto
+          ultimately show ?thesis by simp
+        qed simp_all
+      qed
+    qed
+  qed
+
+subsection \<open>Transfer into the standard model\<close>
+
+interpretation aodv_openproc: openproc paodv opaodv id
+  rewrites "aodv_openproc.initmissing = initmissing"
+  proof -
+    show "openproc paodv opaodv id"
+    proof unfold_locales
+      fix i :: ip
+      have "{(\<sigma>, \<zeta>). (\<sigma> i, \<zeta>) \<in> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V i \<and> (\<forall>j. j \<noteq> i \<longrightarrow> \<sigma> j \<in> fst ` \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V j)} \<subseteq> \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'"
+        unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V'_def
+        proof (rule equalityD1)
+          show "\<And>f p. {(\<sigma>, \<zeta>). (\<sigma> i, \<zeta>) \<in> {(f i, p)} \<and> (\<forall>j. j \<noteq> i
+                      \<longrightarrow> \<sigma> j \<in> fst ` {(f j, p)})} = {(f, p)}"
+            by (rule set_eqI) auto
+        qed
+      thus "{ (\<sigma>, \<zeta>) |\<sigma> \<zeta> s. s \<in> init (paodv i)
+                             \<and> (\<sigma> i, \<zeta>) = id s
+                             \<and> (\<forall>j. j\<noteq>i \<longrightarrow> \<sigma> j \<in> (fst o id) ` init (paodv j)) } \<subseteq> init (opaodv i)"
+        by simp
+    next
+      show "\<forall>j. init (paodv j) \<noteq> {}"
+        unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by simp
+    next
+      fix i s a s' \<sigma> \<sigma>'
+      assume "\<sigma> i = fst (id s)"
+         and "\<sigma>' i = fst (id s')"
+         and "(s, a, s') \<in> trans (paodv i)"
+      then obtain q q' where "s = (\<sigma> i, q)"
+                         and "s' = (\<sigma>' i, q')"
+                         and "((\<sigma> i, q), a, (\<sigma>' i, q')) \<in> trans (paodv i)" 
+         by (cases s, cases s') auto
+      from this(3) have "((\<sigma>, q), a, (\<sigma>', q')) \<in> trans (opaodv i)"
+        by simp (rule open_seqp_action [OF aodv_wf])
+
+      with \<open>s = (\<sigma> i, q)\<close> and \<open>s' = (\<sigma>' i, q')\<close>
+        show "((\<sigma>, snd (id s)), a, (\<sigma>', snd (id s'))) \<in> trans (opaodv i)"
+          by simp
+    qed
+    then interpret opn: openproc paodv opaodv id .
+    have [simp]: "\<And>i. (SOME x. x \<in> (fst o id) ` init (paodv i)) = aodv_init i"
+      unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def by simp
+    hence "\<And>i. openproc.initmissing paodv id i = initmissing i"
+      unfolding opn.initmissing_def opn.someinit_def initmissing_def
+      by (auto split: option.split)
+    thus "openproc.initmissing paodv id = initmissing" ..
+  qed
+
+interpretation aodv_openproc_par_qmsg: openproc_parq paodv opaodv id qmsg
+  rewrites "aodv_openproc_par_qmsg.netglobal = netglobal"
+    and "aodv_openproc_par_qmsg.initmissing = initmissing"
+  proof -
+    show "openproc_parq paodv opaodv id qmsg"
+      by (unfold_locales) simp
+    then interpret opq: openproc_parq paodv opaodv id qmsg .
+
+    have im: "\<And>\<sigma>. openproc.initmissing (\<lambda>i. paodv i \<langle>\<langle> qmsg) (\<lambda>(p, q). (fst (id p), snd (id p), q)) \<sigma>
+                                                                                    = initmissing \<sigma>"
+      unfolding opq.initmissing_def opq.someinit_def initmissing_def
+      unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def \<sigma>\<^sub>Q\<^sub>M\<^sub>S\<^sub>G_def by (clarsimp cong: option.case_cong)
+    thus "openproc.initmissing (\<lambda>i. paodv i \<langle>\<langle> qmsg) (\<lambda>(p, q). (fst (id p), snd (id p), q)) = initmissing"
+      by (rule ext)
+    have "\<And>P \<sigma>. openproc.netglobal (\<lambda>i. paodv i \<langle>\<langle> qmsg) (\<lambda>(p, q). (fst (id p), snd (id p), q)) P \<sigma>
+                                                                                = netglobal P \<sigma>"
+      unfolding opq.netglobal_def netglobal_def opq.initmissing_def initmissing_def opq.someinit_def
+      unfolding \<sigma>\<^sub>A\<^sub>O\<^sub>D\<^sub>V_def \<sigma>\<^sub>Q\<^sub>M\<^sub>S\<^sub>G_def
+      by (clarsimp cong: option.case_cong
+                   simp del: One_nat_def
+                   simp add: fst_initmissing_netgmap_default_aodv_init_netlift
+                                                  [symmetric, unfolded initmissing_def])
+    thus "openproc.netglobal (\<lambda>i. paodv i \<langle>\<langle> qmsg) (\<lambda>(p, q). (fst (id p), snd (id p), q)) = netglobal"
+      by auto
+  qed
+
+lemma net_nhop_quality_increases:
+  assumes "wf_net_tree n"
+  shows "closed (pnet (\<lambda>i. paodv i \<langle>\<langle> qmsg) n) \<TTurnstile> netglobal
+                           (\<lambda>\<sigma>. \<forall>i dip. let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                        in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                                            \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+        (is "_ \<TTurnstile> netglobal (\<lambda>\<sigma>. \<forall>i. ?inv \<sigma> i)")
+  proof -
+    from \<open>wf_net_tree n\<close>
+      have proto: "closed (pnet (\<lambda>i. paodv i \<langle>\<langle> qmsg) n) \<TTurnstile> netglobal (\<lambda>\<sigma>. \<forall>i\<in>net_tree_ips n. \<forall>dip.
+                                            let nhip = the (nhop (rt (\<sigma> i)) dip)
+                                            in dip \<in> vD (rt (\<sigma> i)) \<inter> vD (rt (\<sigma> nhip)) \<and> nhip \<noteq> dip
+                                                \<longrightarrow> (rt (\<sigma> i)) \<sqsubset>\<^bsub>dip\<^esub> (rt (\<sigma> nhip)))"
+        by (rule aodv_openproc_par_qmsg.close_opnet [OF _ onet_nhop_quality_increases])
+    show ?thesis
+    unfolding invariant_def opnet_sos.opnet_tau1
+    proof (rule, simp only: aodv_openproc_par_qmsg.netglobalsimp
+                            fst_initmissing_netgmap_pair_fst, rule allI)
+      fix \<sigma> i
+      assume sr: "\<sigma> \<in> reachable (closed (pnet (\<lambda>i. paodv i \<langle>\<langle> qmsg) n)) TT"
+      hence "\<forall>i\<in>net_tree_ips n. ?inv (fst (initmissing (netgmap fst \<sigma>))) i"
+        by - (drule invariantD [OF proto],
+              simp only: aodv_openproc_par_qmsg.netglobalsimp
+                         fst_initmissing_netgmap_pair_fst)
+      thus "?inv (fst (initmissing (netgmap fst \<sigma>))) i"
+      proof (cases "i\<in>net_tree_ips n")
+        assume "i\<notin>net_tree_ips n"
+        from sr have "\<sigma> \<in> reachable (pnet (\<lambda>i. paodv i \<langle>\<langle> qmsg) n) TT" ..
+        hence "net_ips \<sigma> = net_tree_ips n" ..
+        with \<open>i\<notin>net_tree_ips n\<close> have "i\<notin>net_ips \<sigma>" by simp
+        hence "(fst (initmissing (netgmap fst \<sigma>))) i = aodv_init i"
+          by simp
+        thus ?thesis by simp
+      qed metis
+    qed
+  qed
+
+subsection \<open>Loop freedom of AODV\<close>
+
+theorem aodv_loop_freedom:
+  assumes "wf_net_tree n"
+  shows "closed (pnet (\<lambda>i. paodv i \<langle>\<langle> qmsg) n) \<TTurnstile> netglobal (\<lambda>\<sigma>. \<forall>dip. irrefl ((rt_graph \<sigma> dip)\<^sup>+))"
+  using assms by (rule aodv_openproc_par_qmsg.netglobal_weakenE
+                          [OF net_nhop_quality_increases inv_to_loop_freedom])
+
+end

formal/afp/AODV/variants/b_fwdrreps/B_Aodv_Message.thy

@@ -0,0 +1,74 @@
+(*  Title:       variants/b_fwdrreps/Aodv_Message.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "AODV protocol messages"
+
+theory B_Aodv_Message
+imports B_Fwdrreps
+begin
+
+datatype msg =
+    Rreq nat rreqid ip sqn k ip sqn ip
+  | Rrep nat ip sqn ip ip
+  | Rerr "ip \<rightharpoonup> sqn" ip
+  | Newpkt data ip
+  | Pkt data ip ip
+
+instantiation msg :: msg
+begin
+  definition newpkt_def [simp]: "newpkt \<equiv> \<lambda>(d, dip). Newpkt d dip"
+  definition eq_newpkt_def: "eq_newpkt m \<equiv> case m of Newpkt d dip \<Rightarrow> True | _ \<Rightarrow> False"
+
+  instance by intro_classes (simp add: eq_newpkt_def)
+end
+
+text \<open>The @{type msg} type models the different messages used within AODV.
+      The instantiation as a @{class msg} is a technicality due to the special
+      treatment of @{term newpkt} messages in the AWN SOS rules.
+      This use of classes allows a clean separation of the AWN-specific definitions
+      and these AODV-specific definitions.\<close>
+
+definition rreq :: "nat \<times> rreqid \<times> ip \<times> sqn \<times> k \<times> ip \<times> sqn \<times> ip \<Rightarrow> msg"
+  where "rreq \<equiv> \<lambda>(hops, rreqid, dip, dsn, dsk, oip, osn, sip).
+                    Rreq hops rreqid dip dsn dsk oip osn sip"
+
+lemma rreq_simp [simp]:
+  "rreq(hops, rreqid, dip, dsn, dsk, oip, osn, sip) =  Rreq hops rreqid dip dsn dsk oip osn sip"
+  unfolding rreq_def by simp
+
+definition rrep :: "nat \<times> ip \<times> sqn \<times> ip \<times> ip \<Rightarrow> msg"
+  where "rrep \<equiv> \<lambda>(hops, dip, dsn, oip, sip). Rrep hops dip dsn oip sip"
+
+lemma rrep_simp [simp]:
+  "rrep(hops, dip, dsn, oip, sip) = Rrep hops dip dsn oip sip"
+  unfolding rrep_def by simp
+
+definition rerr :: "(ip \<rightharpoonup> sqn) \<times> ip \<Rightarrow> msg"
+  where "rerr \<equiv> \<lambda>(dests, sip). Rerr dests sip"
+
+lemma rerr_simp [simp]:
+  "rerr(dests, sip) = Rerr dests sip"
+  unfolding rerr_def by simp
+
+lemma not_eq_newpkt_rreq [simp]: "\<not>eq_newpkt (Rreq hops rreqid dip dsn dsk oip osn sip)"
+  unfolding eq_newpkt_def by simp
+
+lemma not_eq_newpkt_rrep [simp]: "\<not>eq_newpkt (Rrep hops dip dsn oip sip)"
+  unfolding eq_newpkt_def by simp
+
+lemma not_eq_newpkt_rerr [simp]: "\<not>eq_newpkt (Rerr dests sip)"
+  unfolding eq_newpkt_def by simp
+
+lemma not_eq_newpkt_pkt [simp]: "\<not>eq_newpkt (Pkt d dip sip)"
+  unfolding eq_newpkt_def by simp
+
+definition pkt :: "data \<times> ip \<times> ip \<Rightarrow> msg"
+  where "pkt \<equiv> \<lambda>(d, dip, sip). Pkt d dip sip"
+
+lemma pkt_simp [simp]:
+  "pkt(d, dip, sip) = Pkt d dip sip"
+  unfolding pkt_def by simp
+
+end

formal/afp/AODV/variants/b_fwdrreps/B_Aodv_Predicates.thy

@@ -0,0 +1,136 @@
+(*  Title:       variants/b_fwdrreps/Aodv_Predicates.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "Invariant assumptions and properties"
+
+theory B_Aodv_Predicates
+imports B_Aodv
+begin
+
+text \<open>Definitions for expression assumptions on incoming messages and properties of
+      outgoing messages.\<close>
+
+abbreviation not_Pkt :: "msg \<Rightarrow> bool"
+where "not_Pkt m \<equiv> case m of Pkt _ _ _ \<Rightarrow> False | _ \<Rightarrow> True"
+
+definition msg_sender :: "msg \<Rightarrow> ip"
+where "msg_sender m \<equiv> case m of Rreq _ _ _ _ _ _ _ ipc \<Rightarrow> ipc
+                              | Rrep _ _ _ _ ipc \<Rightarrow> ipc
+                              | Rerr _ ipc \<Rightarrow> ipc
+                              | Pkt _ _ ipc \<Rightarrow> ipc"
+
+lemma msg_sender_simps [simp]:
+  "\<And>hops rreqid dip dsn dsk oip osn sip.
+                          msg_sender (Rreq hops rreqid dip dsn dsk oip osn sip) = sip"
+  "\<And>hops dip dsn oip sip. msg_sender (Rrep hops dip dsn oip sip) = sip"
+  "\<And>dests sip.            msg_sender (Rerr dests sip) = sip"
+  "\<And>d dip sip.            msg_sender (Pkt d dip sip) = sip"
+  unfolding msg_sender_def by simp_all
+
+definition msg_zhops :: "msg \<Rightarrow> bool"
+where "msg_zhops m \<equiv> case m of
+                                 Rreq hopsc _ dipc _ _ oipc _ sipc \<Rightarrow> hopsc = 0 \<longrightarrow> oipc = sipc
+                               | Rrep hopsc dipc _ _ sipc \<Rightarrow> hopsc = 0 \<longrightarrow> dipc = sipc
+                               | _ \<Rightarrow> True"
+
+lemma msg_zhops_simps [simp]:
+  "\<And>hops rreqid dip dsn dsk oip osn sip.
+           msg_zhops (Rreq hops rreqid dip dsn dsk oip osn sip) = (hops = 0 \<longrightarrow> oip = sip)"
+  "\<And>hops dip dsn oip sip. msg_zhops (Rrep hops dip dsn oip sip) = (hops = 0 \<longrightarrow> dip = sip)"
+  "\<And>dests sip.            msg_zhops (Rerr dests sip)        = True"
+  "\<And>d dip.                msg_zhops (Newpkt d dip)          = True"
+  "\<And>d dip sip.            msg_zhops (Pkt d dip sip)         = True"
+  unfolding msg_zhops_def by simp_all
+
+definition rreq_rrep_sn :: "msg \<Rightarrow> bool"
+where "rreq_rrep_sn m \<equiv> case m of Rreq _ _ _ _ _ _ osnc _ \<Rightarrow> osnc \<ge> 1
+                                | Rrep _ _ dsnc _ _ \<Rightarrow> dsnc \<ge> 1
+                                | _ \<Rightarrow> True"
+
+lemma rreq_rrep_sn_simps [simp]:
+  "\<And>hops rreqid dip dsn dsk oip osn sip.
+           rreq_rrep_sn (Rreq hops rreqid dip dsn dsk oip osn sip) = (osn \<ge> 1)"
+  "\<And>hops dip dsn oip sip. rreq_rrep_sn (Rrep hops dip dsn oip sip) = (dsn \<ge> 1)"
+  "\<And>dests sip.            rreq_rrep_sn (Rerr dests sip) = True"
+  "\<And>d dip.                rreq_rrep_sn (Newpkt d dip)   = True"
+  "\<And>d dip sip.            rreq_rrep_sn (Pkt d dip sip)  = True"
+  unfolding rreq_rrep_sn_def by simp_all
+
+definition rreq_rrep_fresh :: "rt \<Rightarrow> msg \<Rightarrow> bool"
+where "rreq_rrep_fresh crt m \<equiv> case m of Rreq hopsc _ _ _ _ oipc osnc ipcc \<Rightarrow> (ipcc \<noteq> oipc \<longrightarrow>
+                                                oipc\<in>kD(crt) \<and> (sqn crt oipc > osnc
+                                                                \<or> (sqn crt oipc = osnc
+                                                                   \<and> the (dhops crt oipc) \<le> hopsc
+                                                                   \<and> the (flag crt oipc) = val)))
+                                | Rrep hopsc dipc dsnc _ ipcc \<Rightarrow> (ipcc \<noteq> dipc \<longrightarrow> 
+                                                                    dipc\<in>kD(crt)
+                                                                  \<and> sqn crt dipc = dsnc
+                                                                  \<and> the (dhops crt dipc) = hopsc
+                                                                  \<and> the (flag crt dipc) = val)
+                                | _ \<Rightarrow> True"
+
+lemma rreq_rrep_fresh [simp]:
+  "\<And>hops rreqid dip dsn dsk oip osn sip.
+           rreq_rrep_fresh crt (Rreq hops rreqid dip dsn dsk oip osn sip) =
+                               (sip \<noteq> oip \<longrightarrow> oip\<in>kD(crt)
+                                            \<and> (sqn crt oip > osn
+                                               \<or> (sqn crt oip = osn
+                                                  \<and> the (dhops crt oip) \<le> hops
+                                                  \<and> the (flag crt oip) = val)))"
+  "\<And>hops dip dsn oip sip. rreq_rrep_fresh crt (Rrep hops dip dsn oip sip) =
+                               (sip \<noteq> dip \<longrightarrow> dip\<in>kD(crt)
+                                              \<and> sqn crt dip = dsn
+                                              \<and> the (dhops crt dip) = hops
+                                              \<and> the (flag crt dip) = val)"
+  "\<And>dests sip.            rreq_rrep_fresh crt (Rerr dests sip) = True"
+  "\<And>d dip.                rreq_rrep_fresh crt (Newpkt d dip)   = True"
+  "\<And>d dip sip.            rreq_rrep_fresh crt (Pkt d dip sip)  = True"
+  unfolding rreq_rrep_fresh_def by simp_all
+
+definition rerr_invalid :: "rt \<Rightarrow> msg \<Rightarrow> bool"
+where "rerr_invalid crt m \<equiv> case m of Rerr destsc _ \<Rightarrow> (\<forall>ripc\<in>dom(destsc).
+                                            (ripc\<in>iD(crt) \<and> the (destsc ripc) = sqn crt ripc))
+                                | _ \<Rightarrow> True"
+
+lemma rerr_invalid [simp]:
+  "\<And>hops rreqid dip dsn dsk oip osn sip.
+                           rerr_invalid crt (Rreq hops rreqid dip dsn dsk oip osn sip) = True"
+  "\<And>hops dip dsn oip sip. rerr_invalid crt (Rrep hops dip dsn oip sip) = True"
+  "\<And>dests sip.            rerr_invalid crt (Rerr dests sip) = (\<forall>rip\<in>dom(dests).
+                                                 rip\<in>iD(crt) \<and> the (dests rip) = sqn crt rip)"
+  "\<And>d dip.                rerr_invalid crt (Newpkt d dip)   = True"
+  "\<And>d dip sip.            rerr_invalid crt (Pkt d dip sip)  = True"
+  unfolding rerr_invalid_def by simp_all
+
+definition
+  initmissing :: "(nat \<Rightarrow> state option) \<times> 'a \<Rightarrow> (nat \<Rightarrow> state) \<times> 'a"
+where
+  "initmissing \<sigma> = (\<lambda>i. case (fst \<sigma>) i of None \<Rightarrow> aodv_init i | Some s \<Rightarrow> s, snd \<sigma>)"
+
+lemma not_in_net_ips_fst_init_missing [simp]:
+  assumes "i \<notin> net_ips \<sigma>"
+    shows "fst (initmissing (netgmap fst \<sigma>)) i = aodv_init i"
+  using assms unfolding initmissing_def by simp
+
+lemma fst_initmissing_netgmap_pair_fst [simp]:
+  "fst (initmissing (netgmap (\<lambda>(p, q). (fst (id p), snd (id p), q)) s))
+                                               = fst (initmissing (netgmap fst s))"
+  unfolding initmissing_def by auto
+
+text \<open>We introduce a streamlined alternative to @{term initmissing} with @{term netgmap}
+        to simplify invariant statements and thus facilitate their comprehension and
+        presentation.\<close>
+
+lemma fst_initmissing_netgmap_default_aodv_init_netlift:
+  "fst (initmissing (netgmap fst s)) = default aodv_init (netlift fst s)"
+  unfolding initmissing_def default_def
+  by (simp add: fst_netgmap_netlift del: One_nat_def)
+
+definition
+  netglobal :: "((nat \<Rightarrow> state) \<Rightarrow> bool) \<Rightarrow> ((state \<times> 'b) \<times> 'c) net_state \<Rightarrow> bool"
+where
+  "netglobal P \<equiv> (\<lambda>s. P (default aodv_init (netlift fst s)))"
+
+end

formal/afp/AODV/variants/b_fwdrreps/B_Fresher.thy

@@ -0,0 +1,799 @@
+(*  Title:       variants/b_fwdrreps/Fresher.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+*)
+
+section "Quality relations between routes"
+
+theory B_Fresher
+imports B_Aodv_Data
+begin
+
+subsection "Net sequence numbers"
+
+subsubsection "On individual routes"
+
+definition
+  nsqn\<^sub>r :: "r \<Rightarrow> sqn"
+where
+  "nsqn\<^sub>r r \<equiv> if \<pi>\<^sub>4(r) = val \<or> \<pi>\<^sub>2(r) = 0 then \<pi>\<^sub>2(r) else (\<pi>\<^sub>2(r) - 1)"
+
+lemma nsqnr_def':
+  "nsqn\<^sub>r r = (if \<pi>\<^sub>4(r) = inv then \<pi>\<^sub>2(r) - 1 else \<pi>\<^sub>2(r))"
+  unfolding nsqn\<^sub>r_def by simp
+
+lemma nsqn\<^sub>r_zero [simp]:
+  "\<And>dsn dsk flag hops nhip pre. nsqn\<^sub>r (0, dsk, flag, hops, nhip, pre) = 0"
+  unfolding nsqn\<^sub>r_def by clarsimp
+
+lemma nsqn\<^sub>r_val [simp]:
+  "\<And>dsn dsk hops nhip pre. nsqn\<^sub>r (dsn, dsk, val, hops, nhip, pre) = dsn"
+  unfolding nsqn\<^sub>r_def by clarsimp
+
+lemma nsqn\<^sub>r_inv [simp]:
+  "\<And>dsn dsk hops nhip pre. nsqn\<^sub>r (dsn, dsk, inv, hops, nhip, pre) = dsn - 1"
+  unfolding nsqn\<^sub>r_def by clarsimp
+
+lemma nsqn\<^sub>r_lte_dsn [simp]:
+  "\<And>dsn dsk flag hops nhip pre. nsqn\<^sub>r (dsn, dsk, flag, hops, nhip, pre) \<le> dsn"
+  unfolding nsqn\<^sub>r_def by clarsimp
+
+subsubsection "On routes in routing tables"
+
+definition
+  nsqn :: "rt \<Rightarrow> ip \<Rightarrow> sqn"
+where
+  "nsqn \<equiv> \<lambda>rt dip. case \<sigma>\<^bsub>route\<^esub>(rt, dip) of None \<Rightarrow> 0 | Some r \<Rightarrow> nsqn\<^sub>r(r)"
+
+lemma nsqn_sqn_def:
+  "\<And>rt dip. nsqn rt dip = (if flag rt dip = Some val \<or> sqn rt dip = 0
+                            then sqn rt dip else sqn rt dip - 1)"
+  unfolding nsqn_def sqn_def by (clarsimp split: option.split)
+
+lemma not_in_kD_nsqn [simp]:
+  assumes "dip \<notin> kD(rt)"
+  shows "nsqn rt dip = 0"
+  using assms unfolding nsqn_def by simp
+
+lemma kD_nsqn:
+  assumes "dip \<in> kD(rt)"
+    shows "nsqn rt dip = nsqn\<^sub>r(the (\<sigma>\<^bsub>route\<^esub>(rt, dip)))"
+  using assms [THEN kD_Some] unfolding nsqn_def by clarsimp
+
+lemma nsqnr_r_flag_pred [simp, intro]:
+    fixes dsn dsk flag hops nhip pre
+  assumes "P (nsqn\<^sub>r (dsn, dsk, val, hops, nhip, pre))"
+      and "P (nsqn\<^sub>r (dsn, dsk, inv, hops, nhip, pre))"
+    shows "P (nsqn\<^sub>r (dsn, dsk, flag, hops, nhip, pre))"
+  using assms by (cases flag) auto
+
+lemma nsqn\<^sub>r_addpreRT_inv [simp]:
+  "\<And>rt dip npre dip'. dip \<in> kD(rt) \<Longrightarrow>
+   nsqn\<^sub>r (the (the (addpreRT rt dip npre) dip')) = nsqn\<^sub>r (the (rt dip'))"
+  unfolding addpreRT_def nsqn\<^sub>r_def
+  by (frule kD_Some) (clarsimp split: option.split)
+
+lemma sqn_nsqn:
+  "\<And>rt dip. sqn rt dip - 1 \<le> nsqn rt dip"
+  unfolding sqn_def nsqn_def by (clarsimp split: option.split)
+
+lemma nsqn_sqn: "nsqn rt dip \<le> sqn rt dip"
+  unfolding sqn_def nsqn_def by (cases "rt dip") auto
+
+lemma val_nsqn_sqn [elim, simp]:
+  assumes "ip\<in>kD(rt)"
+      and "the (flag rt ip) = val"
+    shows "nsqn rt ip = sqn rt ip"
+  using assms unfolding nsqn_sqn_def by auto
+
+lemma vD_nsqn_sqn [elim, simp]:
+  assumes "ip\<in>vD(rt)"
+    shows "nsqn rt ip = sqn rt ip"
+  proof -
+    from \<open>ip\<in>vD(rt)\<close> have "ip\<in>kD(rt)"
+                      and "the (flag rt ip) = val" by auto
+    thus ?thesis ..
+  qed
+
+lemma inv_nsqn_sqn [elim, simp]:
+  assumes "ip\<in>kD(rt)"
+      and "the (flag rt ip) = inv"
+    shows "nsqn rt ip = sqn rt ip - 1"
+  using assms unfolding nsqn_sqn_def by auto
+
+lemma iD_nsqn_sqn [elim, simp]:
+  assumes "ip\<in>iD(rt)"
+    shows "nsqn rt ip = sqn rt ip - 1"
+  proof -
+    from \<open>ip\<in>iD(rt)\<close> have "ip\<in>kD(rt)"
+                      and "the (flag rt ip) = inv" by auto
+    thus ?thesis ..
+  qed
+
+lemma nsqn_update_changed_kno_val [simp]: "\<And>rt ip dsn dsk hops nhip.
+  rt \<noteq> update rt ip (dsn, kno, val, hops, nhip, {})
+   \<Longrightarrow> nsqn (update rt ip (dsn, kno, val, hops, nhip, {})) ip = dsn"
+  unfolding nsqn\<^sub>r_def update_def
+  by (clarsimp simp: kD_nsqn split: option.split_asm option.split if_split_asm)
+     (metis fun_upd_triv)
+
+lemma nsqn_addpreRT_inv [simp]:
+  "\<And>rt dip npre dip'. dip \<in> kD(rt) \<Longrightarrow>
+   nsqn (the (addpreRT rt dip npre)) dip' = nsqn rt dip'"
+  unfolding addpreRT_def nsqn_def nsqn\<^sub>r_def
+  by (frule kD_Some) (clarsimp split: option.split)
+
+lemma nsqn_update_other [simp]:
+    fixes dsn dsk flag hops dip nhip pre rt ip
+  assumes "dip \<noteq> ip"
+    shows "nsqn (update rt ip (dsn, dsk, flag, hops, nhip, pre)) dip = nsqn rt dip"
+    using assms unfolding nsqn_def
+    by (clarsimp split: option.split)
+
+lemma nsqn_invalidate_eq:
+  assumes "dip \<in> kD(rt)"
+      and "dests dip = Some rsn"
+    shows "nsqn (invalidate rt dests) dip = rsn - 1"
+  using assms
+  proof -
+    from assms obtain dsk hops nhip pre
+      where "invalidate rt dests dip = Some (rsn, dsk, inv, hops, nhip, pre)"
+      unfolding invalidate_def
+      by auto
+    moreover from \<open>dip \<in> kD(rt)\<close> have "dip \<in> kD(invalidate rt dests)" by simp
+    ultimately show ?thesis
+      using \<open>dests dip = Some rsn\<close> by simp
+  qed
+
+lemma nsqn_invalidate_other [simp]:
+  assumes "dip\<in>kD(rt)"
+      and "dip\<notin>dom dests"
+    shows "nsqn (invalidate rt dests) dip = nsqn rt dip"
+  using assms by (clarsimp simp add: kD_nsqn)
+
+subsection "Comparing routes "
+
+definition
+  fresher :: "r \<Rightarrow> r \<Rightarrow> bool" ("(_/ \<sqsubseteq> _)"  [51, 51] 50)
+where
+  "fresher r r' \<equiv> ((nsqn\<^sub>r r < nsqn\<^sub>r r') \<or> (nsqn\<^sub>r r  = nsqn\<^sub>r r' \<and> \<pi>\<^sub>5(r) \<ge> \<pi>\<^sub>5(r')))"
+
+lemma fresherI1 [intro]:
+  assumes "nsqn\<^sub>r r < nsqn\<^sub>r r'"
+    shows "r \<sqsubseteq> r'"
+  unfolding fresher_def using assms by simp
+
+lemma fresherI2 [intro]:
+  assumes "nsqn\<^sub>r r = nsqn\<^sub>r r'"
+      and "\<pi>\<^sub>5(r) \<ge> \<pi>\<^sub>5(r')"
+    shows "r \<sqsubseteq> r'"
+  unfolding fresher_def using assms by simp
+
+lemma fresherI [intro]:
+  assumes "(nsqn\<^sub>r r < nsqn\<^sub>r r') \<or> (nsqn\<^sub>r r  = nsqn\<^sub>r r' \<and> \<pi>\<^sub>5(r) \<ge> \<pi>\<^sub>5(r'))"
+    shows "r \<sqsubseteq> r'"
+  unfolding fresher_def using assms .
+
+lemma fresherE [elim]:
+  assumes "r \<sqsubseteq> r'"
+      and "nsqn\<^sub>r r < nsqn\<^sub>r r' \<Longrightarrow> P r r'"
+      and "nsqn\<^sub>r r  = nsqn\<^sub>r r' \<and> \<pi>\<^sub>5(r) \<ge> \<pi>\<^sub>5(r') \<Longrightarrow> P r r'"
+    shows "P r r'"
+  using assms unfolding fresher_def by auto
+
+lemma fresher_refl [simp]: "r \<sqsubseteq> r"
+  unfolding fresher_def by simp
+
+lemma fresher_trans [elim, trans]:
+  "\<lbrakk> x \<sqsubseteq> y; y \<sqsubseteq> z \<rbrakk> \<Longrightarrow> x \<sqsubseteq> z"
+  unfolding fresher_def by auto
+
+lemma not_fresher_trans [elim, trans]:
+  "\<lbrakk> \<not>(x \<sqsubseteq> y); \<not>(z \<sqsubseteq> x) \<rbrakk> \<Longrightarrow> \<not>(z \<sqsubseteq> y)"
+  unfolding fresher_def by auto
+
+lemma fresher_dsn_flag_hops_const [simp]:
+  fixes dsn dsk dsk' flag hops nhip nhip' pre pre'
+  shows "(dsn, dsk, flag, hops, nhip, pre) \<sqsubseteq> (dsn, dsk', flag, hops, nhip', pre')"
+  unfolding fresher_def by (cases flag) simp_all
+
+lemma addpre_fresher [simp]: "\<And>r npre. r \<sqsubseteq> (addpre r npre)"
+  by clarsimp
+
+subsection "Comparing routing tables "
+
+definition
+  rt_fresher :: "ip \<Rightarrow> rt \<Rightarrow> rt \<Rightarrow> bool"
+where
+  "rt_fresher \<equiv> \<lambda>dip rt rt'. (the (\<sigma>\<^bsub>route\<^esub>(rt, dip))) \<sqsubseteq> (the (\<sigma>\<^bsub>route\<^esub>(rt', dip)))"
+
+abbreviation
+   rt_fresher_syn :: "rt \<Rightarrow> ip \<Rightarrow> rt \<Rightarrow> bool" ("(_/ \<sqsubseteq>\<^bsub>_\<^esub> _)"  [51, 999, 51] 50)
+where
+  "rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2 \<equiv> rt_fresher i rt1 rt2"
+
+lemma rt_fresher_def':
+  "(rt\<^sub>1 \<sqsubseteq>\<^bsub>i\<^esub> rt\<^sub>2) = (nsqn\<^sub>r (the (rt\<^sub>1 i)) < nsqn\<^sub>r (the (rt\<^sub>2 i)) \<or>
+                     nsqn\<^sub>r (the (rt\<^sub>1 i)) = nsqn\<^sub>r (the (rt\<^sub>2 i)) \<and> \<pi>\<^sub>5 (the (rt\<^sub>2 i)) \<le> \<pi>\<^sub>5 (the (rt\<^sub>1 i)))"
+  unfolding rt_fresher_def fresher_def by (rule refl)
+
+lemma single_rt_fresher [intro]:
+  assumes "the (rt1 ip) \<sqsubseteq> the (rt2 ip)"
+    shows "rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2"
+  using assms unfolding rt_fresher_def .
+
+lemma rt_fresher_single [intro]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2"
+    shows "the (rt1 ip) \<sqsubseteq> the (rt2 ip)"
+  using assms unfolding rt_fresher_def .
+
+lemma rt_fresher_def2:
+  assumes "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+    shows "(rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2) = (nsqn rt1 dip < nsqn rt2 dip
+                               \<or> (nsqn rt1 dip = nsqn rt2 dip
+                                   \<and> the (dhops rt1 dip) \<ge> the (dhops rt2 dip)))"
+  using assms unfolding rt_fresher_def fresher_def by (simp add: kD_nsqn proj5_eq_dhops)
+
+lemma rt_fresherI1 [intro]:
+  assumes "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+      and "nsqn rt1 dip < nsqn rt2 dip"
+    shows "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+  unfolding rt_fresher_def2 [OF assms(1-2)] using assms(3) by simp
+
+lemma rt_fresherI2 [intro]:
+  assumes "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+      and "nsqn rt1 dip = nsqn rt2 dip"
+      and "the (dhops rt1 dip) \<ge> the (dhops rt2 dip)"
+    shows "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+  unfolding rt_fresher_def2 [OF assms(1-2)] using assms(3-4) by simp
+
+lemma rt_fresherE [elim]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+      and "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+      and "\<lbrakk> nsqn rt1 dip < nsqn rt2 dip \<rbrakk> \<Longrightarrow> P rt1 rt2 dip"
+      and "\<lbrakk> nsqn rt1 dip = nsqn rt2 dip;
+             the (dhops rt1 dip) \<ge> the (dhops rt2 dip) \<rbrakk> \<Longrightarrow> P rt1 rt2 dip"
+    shows "P rt1 rt2 dip"
+  using assms(1) unfolding rt_fresher_def2 [OF assms(2-3)]
+  using assms(4-5) by auto
+
+lemma rt_fresher_refl [simp]: "rt \<sqsubseteq>\<^bsub>dip\<^esub> rt"
+  unfolding rt_fresher_def by simp
+
+lemma rt_fresher_trans [elim, trans]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+      and "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt3"
+    shows "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt3"
+  using assms unfolding rt_fresher_def by auto
+
+lemma rt_fresher_if_Some [intro!]:
+  assumes "the (rt dip) \<sqsubseteq> r"
+    shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> (\<lambda>ip. if ip = dip then Some r else rt ip)"
+  using assms unfolding rt_fresher_def by simp
+
+definition rt_fresh_as :: "ip \<Rightarrow> rt \<Rightarrow> rt \<Rightarrow> bool"
+where
+  "rt_fresh_as \<equiv> \<lambda>dip rt1 rt2. (rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2) \<and> (rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)"
+
+abbreviation
+   rt_fresh_as_syn :: "rt \<Rightarrow> ip \<Rightarrow> rt \<Rightarrow> bool" ("(_/ \<approx>\<^bsub>_\<^esub> _)"  [51, 999, 51] 50)
+where
+  "rt1 \<approx>\<^bsub>i\<^esub> rt2 \<equiv> rt_fresh_as i rt1 rt2"
+
+lemma rt_fresh_as_refl [simp]: "\<And>rt dip. rt \<approx>\<^bsub>dip\<^esub> rt"
+  unfolding rt_fresh_as_def by simp
+
+lemma rt_fresh_as_trans [simp, intro, trans]:
+  "\<And>rt1 rt2 rt3 dip. \<lbrakk> rt1 \<approx>\<^bsub>dip\<^esub> rt2; rt2 \<approx>\<^bsub>dip\<^esub> rt3 \<rbrakk> \<Longrightarrow> rt1 \<approx>\<^bsub>dip\<^esub> rt3"
+  unfolding rt_fresh_as_def rt_fresher_def
+  by (metis (mono_tags) fresher_trans)
+
+lemma rt_fresh_asI [intro!]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+      and "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1"
+    shows "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+  using assms unfolding rt_fresh_as_def by simp
+
+lemma rt_fresh_as_fresherI [intro]:
+  assumes "dip\<in>kD(rt1)"
+      and "dip\<in>kD(rt2)"
+      and "the (rt1 dip) \<sqsubseteq> the (rt2 dip)"
+      and "the (rt2 dip) \<sqsubseteq> the (rt1 dip)"
+    shows "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+  using assms unfolding rt_fresh_as_def
+  by (clarsimp dest!: single_rt_fresher)
+
+lemma nsqn_rt_fresh_asI:
+  assumes "dip \<in> kD(rt)"
+      and "dip \<in> kD(rt')"
+      and "nsqn rt dip = nsqn rt' dip"
+      and "\<pi>\<^sub>5(the (rt dip)) = \<pi>\<^sub>5(the (rt' dip))"
+    shows "rt \<approx>\<^bsub>dip\<^esub> rt'"
+  proof
+    from assms(1-2,4) have dhops': "the (dhops rt' dip) \<le> the (dhops rt dip)"
+      by (simp add: proj5_eq_dhops)
+    with assms(1-3) show "rt \<sqsubseteq>\<^bsub>dip\<^esub> rt'"
+      by (rule rt_fresherI2)
+  next
+    from assms(1-2,4) have dhops: "the (dhops rt dip) \<le> the (dhops rt' dip)"
+      by (simp add: proj5_eq_dhops)
+    with assms(2,1) assms(3) [symmetric] show "rt' \<sqsubseteq>\<^bsub>dip\<^esub> rt"
+      by (rule rt_fresherI2)
+  qed
+
+lemma rt_fresh_asE [elim]:
+  assumes "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+      and "\<lbrakk> rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2; rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1 \<rbrakk> \<Longrightarrow> P rt1 rt2 dip"
+    shows "P rt1 rt2 dip"
+  using assms unfolding rt_fresh_as_def by simp
+
+lemma rt_fresh_asD1 [dest]:
+  assumes "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    shows "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+  using assms unfolding rt_fresh_as_def by simp
+
+lemma rt_fresh_asD2 [dest]:
+  assumes "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    shows "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1"
+  using assms unfolding rt_fresh_as_def by simp
+
+lemma rt_fresh_as_sym:
+  assumes "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    shows "rt2 \<approx>\<^bsub>dip\<^esub> rt1"
+  using assms unfolding rt_fresh_as_def by simp
+
+lemma not_rt_fresh_asI1 [intro]:
+  assumes "\<not> (rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2)"
+    shows "\<not> (rt1 \<approx>\<^bsub>dip\<^esub> rt2)"
+  proof
+    assume "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    hence "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2" ..
+    with \<open>\<not> (rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2)\<close> show False ..
+  qed
+
+lemma not_rt_fresh_asI2 [intro]:
+  assumes "\<not> (rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)"
+    shows "\<not> (rt1 \<approx>\<^bsub>dip\<^esub> rt2)"
+  proof
+    assume "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    hence "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1" ..
+    with \<open>\<not> (rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)\<close> show False ..
+  qed
+
+lemma not_single_rt_fresher [elim]:
+  assumes "\<not>(the (rt1 ip) \<sqsubseteq> the (rt2 ip))"
+    shows "\<not>(rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2)"
+  proof
+    assume "rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2"
+    hence "the (rt1 ip) \<sqsubseteq> the (rt2 ip)" ..
+    with \<open>\<not>(the (rt1 ip) \<sqsubseteq> the (rt2 ip))\<close> show False ..
+  qed
+
+lemmas not_single_rt_fresh_asI1 [intro] =  not_rt_fresh_asI1 [OF not_single_rt_fresher]
+lemmas not_single_rt_fresh_asI2 [intro] =  not_rt_fresh_asI2 [OF not_single_rt_fresher]
+
+lemma not_rt_fresher_single [elim]:
+  assumes "\<not>(rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2)"
+    shows "\<not>(the (rt1 ip) \<sqsubseteq> the (rt2 ip))"
+  proof
+    assume "the (rt1 ip) \<sqsubseteq> the (rt2 ip)"
+    hence "rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2" ..
+    with \<open>\<not>(rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2)\<close> show False ..
+  qed
+
+lemma rt_fresh_as_nsqnr:
+  assumes "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+      and "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+    shows "nsqn\<^sub>r (the (rt2 dip)) = nsqn\<^sub>r (the (rt1 dip))"
+  using assms(3) unfolding rt_fresh_as_def
+  by (auto simp: rt_fresher_def2 [OF \<open>dip \<in> kD(rt1)\<close> \<open>dip \<in> kD(rt2)\<close>]
+                 rt_fresher_def2 [OF \<open>dip \<in> kD(rt2)\<close> \<open>dip \<in> kD(rt1)\<close>]
+                 kD_nsqn [OF \<open>dip \<in> kD(rt1)\<close>]
+                 kD_nsqn [OF \<open>dip \<in> kD(rt2)\<close>])
+
+lemma rt_fresher_mapupd [intro!]:
+  assumes "dip\<in>kD(rt)"
+      and "the (rt dip) \<sqsubseteq> r"
+    shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> rt(dip \<mapsto> r)"
+  using assms unfolding rt_fresher_def by simp
+
+lemma rt_fresher_map_update_other [intro!]:
+  assumes "dip\<in>kD(rt)"
+      and "dip \<noteq> ip"
+    shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> rt(ip \<mapsto> r)"
+  using assms unfolding rt_fresher_def by simp
+
+lemma rt_fresher_update_other [simp]:
+  assumes inkD: "dip\<in>kD(rt)"
+     and "dip \<noteq> ip"
+   shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> update rt ip r"
+   using assms unfolding update_def
+   by (clarsimp split: option.split) (fastforce)
+
+theorem rt_fresher_update [simp]:
+  assumes "dip\<in>kD(rt)"
+      and "the (dhops rt dip) \<ge> 1"
+      and "update_arg_wf r"
+   shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> update rt ip r"
+  proof (cases "dip = ip")
+    assume "dip \<noteq> ip" with \<open>dip\<in>kD(rt)\<close> show ?thesis
+      by (rule rt_fresher_update_other)
+  next
+    assume "dip = ip"
+
+    from \<open>dip\<in>kD(rt)\<close> obtain dsn\<^sub>n dsk\<^sub>n f\<^sub>n hops\<^sub>n nhip\<^sub>n pre\<^sub>n
+      where rtn [simp]: "the (rt dip) = (dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)"
+      by (metis prod_cases6)
+    with \<open>the (dhops rt dip) \<ge> 1\<close> and \<open>dip\<in>kD(rt)\<close> have "hops\<^sub>n \<ge> 1"
+      by (metis proj5_eq_dhops projs(4))
+    from \<open>dip\<in>kD(rt)\<close> rtn have [simp]: "sqn rt dip = dsn\<^sub>n"
+                           and [simp]: "the (dhops rt dip) = hops\<^sub>n"
+                           and [simp]: "the (flag rt dip) = f\<^sub>n"
+      by (simp add: sqn_def proj5_eq_dhops [symmetric]
+                            proj4_eq_flag [symmetric])+
+
+    from \<open>update_arg_wf r\<close> have "(dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+                                  \<sqsubseteq> the ((update rt dip r) dip)"
+      proof (rule wf_r_cases)
+        fix nhip pre
+        from \<open>hops\<^sub>n \<ge> 1\<close> have "\<And>pre'. (dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+                                        \<sqsubseteq> (dsn\<^sub>n, unk, val, Suc 0, nhip, pre')"
+          unfolding fresher_def sqn_def by (cases f\<^sub>n) auto
+        thus "(dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+               \<sqsubseteq> the (update rt dip (0, unk, val, Suc 0, nhip, pre) dip)"
+          using \<open>dip\<in>kD(rt)\<close> by - (rule update_cases_kD, simp_all)
+      next
+        fix dsn :: sqn and hops nhip pre
+        assume "0 < dsn"
+        show "(dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+               \<sqsubseteq> the (update rt dip (dsn, kno, val, hops, nhip, pre) dip)"
+        proof (rule update_cases_kD [OF _ \<open>dip\<in>kD(rt)\<close>], simp_all add: \<open>0 < dsn\<close>)
+          assume "dsn\<^sub>n < dsn"
+            thus "(dsn\<^sub>n, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+                   \<sqsubseteq> (dsn, kno, val, hops, nhip, pre \<union> pre\<^sub>n)"
+              unfolding fresher_def by auto
+        next
+          assume "dsn\<^sub>n = dsn"
+             and "hops < hops\<^sub>n"
+            thus "(dsn, dsk\<^sub>n, f\<^sub>n, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+                   \<sqsubseteq> (dsn, kno, val, hops, nhip, pre \<union> pre\<^sub>n)"
+              unfolding fresher_def nsqn\<^sub>r_def by simp
+        next
+          assume "dsn\<^sub>n = dsn"
+            with \<open>0 < dsn\<close>
+            show "(dsn, dsk\<^sub>n, inv, hops\<^sub>n, nhip\<^sub>n, pre\<^sub>n)
+                   \<sqsubseteq> (dsn, kno, val, hops, nhip, pre \<union> pre\<^sub>n)"
+              unfolding fresher_def by simp
+        qed
+      qed
+    hence "rt \<sqsubseteq>\<^bsub>dip\<^esub> update rt dip r"
+      by - (rule single_rt_fresher, simp)
+    with \<open>dip = ip\<close> show ?thesis by simp
+  qed
+
+theorem rt_fresher_invalidate [simp]:
+  assumes "dip\<in>kD(rt)"
+      and indests: "\<forall>rip\<in>dom(dests). rip\<in>vD(rt) \<and> sqn rt rip < the (dests rip)"
+    shows "rt \<sqsubseteq>\<^bsub>dip\<^esub> invalidate rt dests"
+  proof (cases "dip\<in>dom(dests)")
+    assume "dip\<notin>dom(dests)"
+      thus ?thesis using \<open>dip\<in>kD(rt)\<close>
+      by - (rule single_rt_fresher, simp)
+  next
+    assume "dip\<in>dom(dests)"
+    moreover with indests have "dip\<in>vD(rt)"
+                           and "sqn rt dip < the (dests dip)"
+      by auto
+    ultimately show ?thesis
+      unfolding invalidate_def sqn_def
+      by - (rule single_rt_fresher, auto simp: fresher_def)
+  qed
+
+lemma nsqn\<^sub>r_invalidate [simp]:
+  assumes "dip\<in>kD(rt)"
+      and "dip\<in>dom(dests)"
+    shows "nsqn\<^sub>r (the (invalidate rt dests dip)) = the (dests dip) - 1"
+  using assms unfolding invalidate_def by auto
+
+lemma rt_fresh_as_inc_invalidate [simp]:
+  assumes "dip\<in>kD(rt)"
+      and "\<forall>rip\<in>dom(dests). rip\<in>vD(rt) \<and> the (dests rip) = inc (sqn rt rip)"
+    shows "rt \<approx>\<^bsub>dip\<^esub> invalidate rt dests"
+  proof (cases "dip\<in>dom(dests)")
+    assume "dip\<notin>dom(dests)"
+    with \<open>dip\<in>kD(rt)\<close> have "dip\<in>kD(invalidate rt dests)"
+      by simp
+    with \<open>dip\<in>kD(rt)\<close> show ?thesis
+      by rule (simp_all add: \<open>dip\<notin>dom(dests)\<close>)
+  next
+    assume "dip\<in>dom(dests)"
+    with assms(2) have "dip\<in>vD(rt)"
+                  and "the (dests dip) = inc (sqn rt dip)" by auto
+    from \<open>dip\<in>vD(rt)\<close> have "dip\<in>kD(rt)" by simp
+    moreover then have "dip\<in>kD(invalidate rt dests)" by simp
+    ultimately show ?thesis
+    proof (rule nsqn_rt_fresh_asI)
+      from \<open>dip\<in>vD(rt)\<close> have "nsqn rt dip = sqn rt dip" by simp
+      also have "sqn rt dip = nsqn\<^sub>r (the (invalidate rt dests dip))"
+      proof -
+        from \<open>dip\<in>kD(rt)\<close> have "nsqn\<^sub>r (the (invalidate rt dests dip)) = the (dests dip) - 1"
+          using \<open>dip\<in>dom(dests)\<close> by (rule nsqn\<^sub>r_invalidate)
+        with \<open>the (dests dip) = inc (sqn rt dip)\<close>
+          show "sqn rt dip = nsqn\<^sub>r (the (invalidate rt dests dip))" by simp
+      qed
+      also from \<open>dip\<in>kD(invalidate rt dests)\<close>
+        have "nsqn\<^sub>r (the (invalidate rt dests dip)) = nsqn (invalidate rt dests) dip"
+          by (simp add: kD_nsqn)
+      finally show "nsqn rt dip = nsqn (invalidate rt dests) dip" .
+    qed simp
+  qed
+
+lemmas rt_fresher_inc_invalidate [simp] = rt_fresh_as_inc_invalidate [THEN rt_fresh_asD1]
+
+lemma rt_fresh_as_addpreRT [simp]:
+  assumes "ip\<in>kD(rt)"
+    shows "rt \<approx>\<^bsub>dip\<^esub> the (addpreRT rt ip npre)"
+  using assms [THEN kD_Some] by (auto simp: addpreRT_def)
+
+lemmas rt_fresher_addpreRT [simp] = rt_fresh_as_addpreRT [THEN rt_fresh_asD1]
+
+subsection "Strictly comparing routing tables "
+
+definition rt_strictly_fresher :: "ip \<Rightarrow> rt \<Rightarrow> rt \<Rightarrow> bool"
+where
+  "rt_strictly_fresher \<equiv> \<lambda>dip rt1 rt2. (rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2) \<and> \<not>(rt1 \<approx>\<^bsub>dip\<^esub> rt2)"
+
+abbreviation
+   rt_strictly_fresher_syn :: "rt \<Rightarrow> ip \<Rightarrow> rt \<Rightarrow> bool" ("(_/ \<sqsubset>\<^bsub>_\<^esub> _)"  [51, 999, 51] 50)
+where
+  "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2 \<equiv> rt_strictly_fresher i rt1 rt2"
+
+lemma rt_strictly_fresher_def'':
+  "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2 = ((rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2) \<and> \<not>(rt2 \<sqsubseteq>\<^bsub>i\<^esub> rt1))"
+  unfolding rt_strictly_fresher_def rt_fresh_as_def by auto
+
+lemma rt_strictly_fresherI' [intro]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2"
+      and "\<not>(rt2 \<sqsubseteq>\<^bsub>i\<^esub> rt1)"
+    shows "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2"
+  using assms unfolding rt_strictly_fresher_def'' by simp
+
+lemma rt_strictly_fresherE' [elim]:
+  assumes "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2"
+      and "\<lbrakk> rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2; \<not>(rt2 \<sqsubseteq>\<^bsub>i\<^esub> rt1) \<rbrakk> \<Longrightarrow> P rt1 rt2 i"
+    shows "P rt1 rt2 i"
+  using assms unfolding rt_strictly_fresher_def'' by simp
+
+lemma rt_strictly_fresherI [intro]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2"
+      and "\<not>(rt1 \<approx>\<^bsub>i\<^esub> rt2)"
+    shows "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2"
+  unfolding rt_strictly_fresher_def using assms ..
+
+lemmas rt_strictly_fresher_singleI [elim] = rt_strictly_fresherI [OF single_rt_fresher]
+
+lemma rt_strictly_fresherE [elim]:
+  assumes "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2"
+      and "\<lbrakk> rt1 \<sqsubseteq>\<^bsub>i\<^esub> rt2; \<not>(rt1 \<approx>\<^bsub>i\<^esub> rt2) \<rbrakk> \<Longrightarrow> P rt1 rt2 i"
+    shows "P rt1 rt2 i"
+  using assms(1) unfolding rt_strictly_fresher_def
+  by rule (erule(1) assms(2))
+
+lemma rt_strictly_fresher_def':
+  "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2 =
+     (nsqn\<^sub>r (the (rt1 i)) < nsqn\<^sub>r (the (rt2 i))
+       \<or> (nsqn\<^sub>r (the (rt1 i)) = nsqn\<^sub>r (the (rt2 i)) \<and> \<pi>\<^sub>5(the (rt1 i)) > \<pi>\<^sub>5(the (rt2 i))))"
+  unfolding rt_strictly_fresher_def'' rt_fresher_def fresher_def by auto
+
+lemma rt_strictly_fresher_fresherD [dest]:
+  assumes "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2"
+    shows "the (rt1 dip) \<sqsubseteq> the (rt2 dip)"
+  using assms unfolding rt_strictly_fresher_def rt_fresher_def by auto
+
+lemma rt_strictly_fresher_not_fresh_asD [dest]:
+  assumes "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2"
+    shows "\<not> rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+  using assms unfolding rt_strictly_fresher_def by auto
+
+lemma rt_strictly_fresher_trans [elim, trans]:
+  assumes "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2"
+      and "rt2 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+    shows "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+  using assms proof -
+    from \<open>rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2\<close> obtain "the (rt1 dip) \<sqsubseteq> the (rt2 dip)" by auto
+    also from \<open>rt2 \<sqsubset>\<^bsub>dip\<^esub> rt3\<close> obtain "the (rt2 dip) \<sqsubseteq> the (rt3 dip)" by auto
+    finally have "the (rt1 dip) \<sqsubseteq> the (rt3 dip)" .
+
+    moreover have "\<not> (rt1 \<approx>\<^bsub>dip\<^esub> rt3)"
+    proof -    
+      from \<open>rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2\<close> obtain "\<not>(the (rt2 dip) \<sqsubseteq> the (rt1 dip))" by auto
+      also from \<open>rt2 \<sqsubset>\<^bsub>dip\<^esub> rt3\<close> obtain "\<not>(the (rt3 dip) \<sqsubseteq> the (rt2 dip))" by auto
+      finally have "\<not>(the (rt3 dip) \<sqsubseteq> the (rt1 dip))" .
+      thus ?thesis ..
+    qed
+    ultimately show "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3" ..
+ qed
+
+lemma rt_strictly_fresher_irefl [simp]: "\<not> (rt \<sqsubset>\<^bsub>dip\<^esub> rt)"
+  unfolding rt_strictly_fresher_def
+  by clarsimp
+
+lemma rt_fresher_trans_rt_strictly_fresher [elim, trans]:
+  assumes "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2"
+      and "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt3"
+    shows "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+  proof -
+    from \<open>rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2\<close> have "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+                           and "\<not>(rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)"
+      unfolding rt_strictly_fresher_def'' by auto
+    from this(1) and \<open>rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt3\<close> have "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt3" ..
+
+    moreover from \<open>\<not>(rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)\<close> have "\<not>(rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)"
+      proof (rule contrapos_nn)
+        assume "rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt1"
+        with \<open>rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt3\<close> show "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1" ..
+      qed
+
+    ultimately show "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+      unfolding rt_strictly_fresher_def'' by auto
+  qed
+
+lemma rt_fresher_trans_rt_strictly_fresher' [elim, trans]:
+  assumes "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2"
+      and "rt2 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+    shows "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+  proof -
+    from \<open>rt2 \<sqsubset>\<^bsub>dip\<^esub> rt3\<close> have "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt3"
+                           and "\<not>(rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt2)"
+      unfolding rt_strictly_fresher_def'' by auto
+    from \<open>rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2\<close> and this(1) have "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt3" ..
+
+    moreover from \<open>\<not>(rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt2)\<close> have "\<not>(rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt1)"
+      proof (rule contrapos_nn)
+        assume "rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt1"
+        thus "rt3 \<sqsubseteq>\<^bsub>dip\<^esub> rt2" using \<open>rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2\<close> ..
+      qed
+
+    ultimately show "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt3"
+      unfolding rt_strictly_fresher_def'' by auto
+  qed
+
+lemma rt_fresher_imp_nsqn_le:
+  assumes "rt1 \<sqsubseteq>\<^bsub>ip\<^esub> rt2"
+      and "ip \<in> kD rt1"
+      and "ip \<in> kD rt2"
+    shows "nsqn rt1 ip \<le> nsqn rt2 ip"
+  using assms(1)
+  by (auto simp add: rt_fresher_def2 [OF assms(2-3)])
+
+lemma rt_strictly_fresher_ltI [intro]:
+  assumes "dip \<in> kD(rt1)"
+      and "dip \<in> kD(rt2)"
+      and "nsqn rt1 dip < nsqn rt2 dip"
+    shows "rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2"
+  proof
+    from assms show "rt1 \<sqsubseteq>\<^bsub>dip\<^esub> rt2" ..
+  next
+    show "\<not>(rt1 \<approx>\<^bsub>dip\<^esub> rt2)"
+    proof
+      assume "rt1 \<approx>\<^bsub>dip\<^esub> rt2"
+      hence "rt2 \<sqsubseteq>\<^bsub>dip\<^esub> rt1" ..
+      hence "nsqn rt2 dip \<le> nsqn rt1 dip"
+        using \<open>dip \<in> kD(rt2)\<close> \<open>dip \<in> kD(rt1)\<close>
+        by (rule rt_fresher_imp_nsqn_le)
+      with \<open>nsqn rt1 dip < nsqn rt2 dip\<close> show "False"
+        by simp
+    qed
+  qed
+
+lemma rt_strictly_fresher_eqI [intro]:
+  assumes "i\<in>kD(rt1)"
+      and "i\<in>kD(rt2)"
+      and "nsqn rt1 i = nsqn rt2 i"
+      and "\<pi>\<^sub>5(the (rt2 i)) < \<pi>\<^sub>5(the (rt1 i))"
+    shows "rt1 \<sqsubset>\<^bsub>i\<^esub> rt2"
+  using assms unfolding rt_strictly_fresher_def' by (auto simp add: kD_nsqn)
+
+lemma invalidate_rtsf_left [simp]:
+  "\<And>dests dip rt rt'. dests dip = None \<Longrightarrow> (invalidate rt dests \<sqsubset>\<^bsub>dip\<^esub> rt') = (rt \<sqsubset>\<^bsub>dip\<^esub> rt')"
+  unfolding invalidate_def rt_strictly_fresher_def'
+  by (rule iffI) (auto split: option.split_asm)
+
+lemma vD_invalidate_rt_strictly_fresher [simp]:
+  assumes "dip \<in> vD(invalidate rt1 dests)"
+    shows "(invalidate rt1 dests \<sqsubset>\<^bsub>dip\<^esub> rt2) = (rt1 \<sqsubset>\<^bsub>dip\<^esub> rt2)"
+  proof (cases "dip \<in> dom(dests)")
+    assume "dip \<in> dom(dests)"
+    hence "dip \<notin> vD(invalidate rt1 dests)"
+      unfolding invalidate_def vD_def
+      by clarsimp (metis assms option.simps(3) vD_invalidate_vD_not_dests)
+    with \<open>dip \<in> vD(invalidate rt1 dests)\<close> show ?thesis by simp
+  next
+    assume "dip \<notin> dom(dests)"
+    hence "dests dip = None" by auto
+    moreover with \<open>dip \<in> vD(invalidate rt1 dests)\<close> have "dip \<in> vD(rt1)"
+      unfolding invalidate_def vD_def
+      by clarsimp (metis (opaque_lifting, no_types) assms vD_Some vD_invalidate_vD_not_dests)
+    ultimately show ?thesis
+      unfolding invalidate_def rt_strictly_fresher_def' by auto
+  qed
+
+lemma rt_strictly_fresher_update_other [elim!]:
+  "\<And>dip ip rt r rt'. \<lbrakk> dip \<noteq> ip; rt \<sqsubset>\<^bsub>dip\<^esub> rt' \<rbrakk> \<Longrightarrow> update rt ip r \<sqsubset>\<^bsub>dip\<^esub> rt'"
+  unfolding rt_strictly_fresher_def' by clarsimp
+
+lemma addpreRT_strictly_fresher [simp]:
+  assumes "dip \<in> kD(rt)"
+    shows "(the (addpreRT rt dip npre) \<sqsubset>\<^bsub>ip\<^esub> rt2) = (rt \<sqsubset>\<^bsub>ip\<^esub> rt2)"
+  using assms unfolding rt_strictly_fresher_def' by clarsimp
+
+lemma lt_sqn_imp_update_strictly_fresher:
+  assumes "dip \<in> vD (rt2 nhip)"
+      and  *: "osn < sqn (rt2 nhip) dip"
+      and **: "rt \<noteq> update rt dip (osn, kno, val, hops, nhip, {})"
+    shows "update rt dip (osn, kno, val, hops, nhip, {}) \<sqsubset>\<^bsub>dip\<^esub> rt2 nhip"
+  unfolding rt_strictly_fresher_def'
+  proof (rule disjI1)
+    from ** have "nsqn (update rt dip (osn, kno, val, hops, nhip, {})) dip = osn"
+      by (rule nsqn_update_changed_kno_val)
+    with \<open>dip\<in>vD(rt2 nhip)\<close>
+      have "nsqn\<^sub>r (the (update rt dip (osn, kno, val, hops, nhip, {}) dip)) = osn"
+        by (simp add: kD_nsqn)
+    also have "osn < sqn (rt2 nhip) dip" by (rule *)
+    also have "sqn (rt2 nhip) dip = nsqn\<^sub>r (the (rt2 nhip dip))"
+      unfolding nsqn\<^sub>r_def using \<open>dip \<in> vD (rt2 nhip)\<close>
+      by - (metis vD_flag_val proj2_eq_sqn proj4_eq_flag vD_iD_gives_kD(1))
+    finally show "nsqn\<^sub>r (the (update rt dip (osn, kno, val, hops, nhip, {}) dip))
+                                                              < nsqn\<^sub>r (the (rt2 nhip dip))" .
+  qed
+
+lemma dhops_le_hops_imp_update_strictly_fresher:
+  assumes "dip \<in> vD(rt2 nhip)"
+      and sqn: "sqn (rt2 nhip) dip = osn"
+      and hop: "the (dhops (rt2 nhip) dip) \<le> hops"
+      and **: "rt \<noteq> update rt dip (osn, kno, val, Suc hops, nhip, {})"
+    shows "update rt dip (osn, kno, val, Suc hops, nhip, {}) \<sqsubset>\<^bsub>dip\<^esub> rt2 nhip"
+  unfolding rt_strictly_fresher_def'
+  proof (rule disjI2, rule conjI)
+    from ** have "nsqn (update rt dip (osn, kno, val, Suc hops, nhip, {})) dip = osn"
+      by (rule nsqn_update_changed_kno_val)
+    with \<open>dip\<in>vD(rt2 nhip)\<close>
+      have "nsqn\<^sub>r (the (update rt dip (osn, kno, val, Suc hops, nhip, {}) dip)) = osn"
+        by (simp add: kD_nsqn)
+    also have "osn = sqn (rt2 nhip) dip" by (rule sqn [symmetric])
+    also have "sqn (rt2 nhip) dip = nsqn\<^sub>r (the (rt2 nhip dip))"
+      unfolding nsqn\<^sub>r_def using \<open>dip \<in> vD(rt2 nhip)\<close>
+      by - (metis vD_flag_val proj2_eq_sqn proj4_eq_flag vD_iD_gives_kD(1))
+    finally show "nsqn\<^sub>r (the (update rt dip (osn, kno, val, Suc hops, nhip, {}) dip))
+                                                              = nsqn\<^sub>r (the (rt2 nhip dip))" .
+  next
+    have "the (dhops (rt2 nhip) dip) \<le> hops" by (rule hop)
+    also have "hops < hops + 1" by simp
+    also have "hops + 1 = the (dhops (update rt dip (osn, kno, val, Suc hops, nhip, {})) dip)"
+      using ** by simp
+    finally have "the (dhops (rt2 nhip) dip)
+                        < the (dhops (update rt dip (osn, kno, val, Suc hops, nhip, {})) dip)" .
+    thus "\<pi>\<^sub>5 (the (rt2 nhip dip)) < \<pi>\<^sub>5 (the (update rt dip (osn, kno, val, Suc hops, nhip, {}) dip))"
+      using \<open>dip \<in> vD(rt2 nhip)\<close> by (simp add: proj5_eq_dhops)
+  qed
+
+lemma nsqn_invalidate:
+  assumes "dip \<in> kD(rt)"
+      and "\<forall>ip\<in>dom(dests). ip \<in> vD(rt) \<and> the (dests ip) = inc (sqn rt ip)"
+    shows "nsqn (invalidate rt dests) dip = nsqn rt dip"
+  proof -
+    from \<open>dip \<in> kD(rt)\<close> have "dip \<in> kD(invalidate rt dests)" by simp
+
+    from assms have "rt \<approx>\<^bsub>dip\<^esub> invalidate rt dests"
+      by (rule rt_fresh_as_inc_invalidate)
+    with \<open>dip \<in> kD(rt)\<close> \<open>dip \<in> kD(invalidate rt dests)\<close> show ?thesis
+      by (simp add: kD_nsqn del: invalidate_kD_inv)
+         (erule(2) rt_fresh_as_nsqnr)
+  qed
+
+end

formal/afp/AODV/variants/b_fwdrreps/B_Fwdrreps.thy

@@ -0,0 +1,33 @@
+(*  Title:       variants/b_fwdrreps/B_Fwdrreps.thy
+    License:     BSD 2-Clause. See LICENSE.
+    Author:      Timothy Bourke, Inria
+    Author:      Peter Höfner, NICTA
+*)
+
+theory %invisible B_Fwdrreps
+imports "../../Aodv_Basic"
+begin
+
+chapter "Variant B: Forwarding the Route Reply"
+
+text \<open>
+  Explanation~\cite[\textsection 10.2]{FehnkerEtAl:AWN:2013}:
+  In AODV's route discovery process, a RREP message from the destination 
+  node is unicast back along a route towards the originator of the RREQ 
+  message. Every intermediate node on the selected route will process the 
+  RREP message and, in most cases, forward it towards the originator node. 
+  However, there is a possibility that the RREP message is discarded at an 
+  intermediate node, which results in the originator node not receiving a 
+  reply. The discarding of the RREP message is due to the RFC specification 
+  of AODV~\cite{RFC3561} stating that an intermediate node only forwards the 
+  RREP message if it is not the originator node and it has created or 
+  updated a routing table entry to the destination node described in the 
+  RREP message. The latter requirement means that if a valid routing table 
+  entry to the destination node already exists, and is not updated when 
+  processing the RREP message, then the intermediate node will not forward 
+  the message. A solution to this problem is to require intermediate nodes 
+  to forward all RREP messages that they receive.
+\<close>
+
+end %invisible
+